
Vorwort 

Die Erscheinung des REXXikons ist jetzt 14 Jahre her und da es immer noch kein vergleichbares 
Gesamtwerk gibt, habe ich mich nach langem Zögern entschlossen, eine aktualisierte und erweiterte 
Ausgabe zu liefern. Das vorliegende Exemplar basiert auf z/OS 2.4. Ergänzend gibt es Kapitel über 
die API zu SDSF und DSNREXX, der DB2-Schnittstelle unter TSO. 

Nein, auch dieses soll keine wissenschaftliche Abhandlung sein. Der Antrieb ist immer noch, ein 
Buch aus der Praxis für die Praxis. 

Mittlerweile wird der Mainframe nicht mehr als antiquierter Moloch angesehen, dessen Technik ver-
altet ist. In vielen Bereichen von Banken, Versicherungen und Industrie ist der Mainframe mangels 
Alternativen wieder gesetzt. Ab einer gewissen Unternehmensgröße ist die Abschaffung nur zu ma-
chen, wenn Geld und Vernunft keinerlei Rolle spielen – auch wenn dies zugegebener maßen meine 
ganz persönliche Anschauung ist. 

Seit einiger Zeit gibt es auch wieder größere Aktivitäten im Ausbildungsbereich. Der Mainframe ist 
nicht mehr allen Universitäten fremd und auch einige Unternehmen haben entdeckt, dass Fachkräfte 
nicht auf den Bäumen wachsen. Man tut wieder etwas im Ausbildungsbereich. 

Nach nunmehr 47 Jahren Erfahrung mit dem Großrechner erfreue ich mich noch immer an meinem 
geliebten Mainframe. Mittlerweile ist auch eingetroffen, wovon ich auch schon im zarten Alter von 
Ende Vierzig überzeugt war: Ich bin vor dem Mainframe in Rente gegangen. 

Danksagung an… 

…alle Seminarteilnehmer, welche in der Vergangenheit unter mir leiden mussten. Ich bin immer 
wieder erstaunt, welch unterschiedliche Denkansätze es gibt. Das Eine oder Andere findet sich in 
den Beispielen wieder. 

…die NewsGroups TSOREXX, ISPF-L und IBM-MAIN. Eine Art OpenSource-Gemeinde. Gut finde 
ich auch, dass ihr nicht jedem die Arbeit abnehmt, nur weil er zu faul ist, im Manual nachzulesen. 

…www.cbttape.org, der MVS-Freeware-Seite. Oft habe ich mich gefragt, wie ich meine Zeit besser 
nutze: Kodieren oder die Lösung dort suchen. 

…Yves Colliard, von dem ich das eine oder andere Beispiel "gestohlen" habe. Ein Besuch auf seiner 
Homepage lohnt sich. Es gibt viele praktische Beispiele für den täglichen Gebrauch zum kostenlosen 
Download (www.ycos.de).  

…Raimund Süss, der nach dem letzten Buch einige Anregungen zu Themen, welche ihm "gefehlt" 
haben, gegeben hat. Hallo Raimund: I did my very best. Danke für die Vorschläge. 

…Christian Zitzmann, der das Skript Probe gelesen und auf fachliche Fehler hin überprüft hat. Chris-
tian, wir haben viel voneinander gelernt. Bleib am Ball. Der Host wird Dich noch lange sehr gut 
ernähren. Und besten Dank für Dein permanentes "wieso?". Auch wenn es manchmal nervig war, 
hat es mich doch gezwungen, über alles, was ich Dir zu erklären versuchte, nochmal nachzudenken. 

…last but not least, meiner Frau Claudia, die mir immer dann weitergeholfen hat, wenn ich beim 
Umgang mit meinem Textverarbeitungsprogramm kurz vor einem Nervenzusammenbruch stand. 

  

http://www.ycos.de/

Die Zielgruppe 

Ich möchte alle erreichen. Dennoch: Eine gewisse Grundlage sollte vorhanden sein. Das Dokument 
richtet sich an z/OS-Nutzer, welche mit REXX einfache oder komplexe Anwendungen unter TSO 
entwickeln wollen, und bereits Erfahrung in einer Programmiersprache gesammelt haben. Es han-
delt sich um kein Lehrbuch, sondern vielmehr um die praktische Hilfe am Arbeitsplatz – ein Nach-
schlagewerk mit einer Vielzahl von Beispielen für die tägliche Arbeit. Es lebt von der Praxis, nicht 
von der Theorie. 

Die Beispiele sollen Hilfe bei der Entwicklung geben. REXX-native oder Dialoge, Helfer im Alltag 
durch Edit Makros. JCL-Beispiele für Batchverarbeitung, wo sie sinnvoll und vernünftiger als eine 
Foreground-Verarbeitung ist. Und vieles mehr… 

Willkommen sind auch alle, welche die Effizienz ihrer Leistung steigern möchten oder die Outputs 
ihrer Anwendung per Programm überprüfen wollen – oft reichen hierzu ein paar Lines of Code und 
REXX ist oft schneller kodiert, als Cobol compiled. 

Für diejenigen, welche schon alles wissen, schlägt mein Herz natürlich auch. Allerdings habe ich 
eine Bitte: Der Host lebt! Wann habt ihr euch das letzte Mal mit den Neuerungen auseinanderge-
setzt? Ich möchte nicht den Eindruck erwecken, alles zu wissen. Es hat mich aber oft erstaunt, wie 
häufig Neuerungen an Anwendern vorbeigehen und wie leicht man auf einem bestimmten Knowhow 
stehen bleibt. Es geht ja auch immer irgendwie. Wenn ein Problem nicht sofort lösbar ist, umfliegt 
man es. Für euch versuche ich, alle Fragen zu beantworten, die ich mir oft selbst gestellt habe. 

Versprechen 

Wenn Du für das Eine oder andere Unterstützung brauchst und in Ihrer unmittelbaren Umgebung 
keine findest, schreib mir einfach: rexxikon@spgmbh.de. Ich hätte gerne Deinen Code, eine kurze 
Beschreibung der Problemstellung und was Deiner Meinung nach falsch läuft. Du bekommst sicher 
Antwort. Versprochen. 

Die Struktur des Rexxikons 

Das wichtigste vorab: Es sind 2 Teile geworden. Der erste, das "Kochbuch", ist all jenen gewidmet, 
die sich erst einen Überblick über die vielfältigen Möglichkeinen von REXX und seinen Randberei-
chen verschaffen wollen. Der zweite enthält die ausgelagerte Syntaxbeschreibung aller Befehle und 
Funktionen und sollte in erster Linie eine Hilfe bei der täglichen Arbeit sein.  

Der Grund für die Zweiteilung: Es sollten nicht mehr als 2 Kilo Buch herauskommen, sofern es ge-
druckt werden sollte. Aus diesem Grund ist das Format jetzt auch DIN-A4. 

Generell sollen hier unterschiedliche Anforderungen erfüllt werden. In den verschiedenen Kapiteln 
soll für jeden schnell auffindbar das angeboten werden, was gerade gebraucht wird. 

Band-1, Kochbuch 

Kapitel 1 widmet sich der Erklärung der allgemeinen Dinge. Die Sprache REXX wird erläutert, die 
Umgebung erklärt. Es werden diverse Beispiele aus der täglichen Praxis aufgeführt.  

Kapitel 2 setzt sich mit Edit Makros auseinander. In der täglichen Praxis sind REXX-native Anwen-
dungen eher selten. Ich frage mich oft, wie jemand vernünftig im Editor arbeiten kann, ohne die 
vielen kleinen wiederkehrenden Dinge des Lebens über ein Makro abzudecken. Schnell mal eine 
Datei sortieren, den Cursor auf einen Dateinamen zu stellen, um in diese reinzuschauen oder einen 
Auto-Vervollständiger zu benutzen. Im Internet sind wir dies gewohnt und empfinden es als prak-
tisch. Warum nicht auch im z/OS? 

Kapitel 3 liefert einen Überblick über die LM-Services. Natürlich sind diese heute nicht mehr von so 
großer Bedeutung, wie in den 80er Jahren. Spanned Records und Lademodule können heute auch 
mit EXECIO gelesen werden. Funktionen wie Löschen oder Umbenennen von Membern können 
auch über TSO-Befehle ausgeführt werden. Aber was Komfort und Performance angeht, sind die 
LM-Services ihren Pendants im TSO weit überlegen. 

Kapitel 4 beschreibt ISPF (Dialog Manager). Neben der Erklärung der Services wird das Wirken des 
Dialog Managers anhand einer kleinen Anwendung demonstriert, welche zeigt, wie die einzelnen 
Komponenten zusammenspielen. 

Kapitel 5 anhand einiger Anwendungen, wie die Zugriffe auf SDSF aus REXX möglich sind. Es ist 
verblüffend an wie vielen Stellen heute noch Coding zu finden ist, wie man es vor 25 Jahren benut-
zen musste. Heute gehen die Dinge viel besser und einfacher. Übrigens: Es ist immer nur solange 
alles kompliziert, bis man verstanden hat, wie es funktioniert. 

Kapitel 6 zeigt den Umgang mit DSNREXX, der REXX/TSO-Schnittstelle zu DB2. Hier findet ihr eine 
kleine Anwendung, welche in einer bestehenden Datenbank einen Tablespace und eine Tabelle 
erzeugt und initial aus einer Datei befüllt. In einer Dialoganwendung wird veranschaulicht, wie die 
Daten aus DB2 über eine temporäre ISPF-Tabelle angezeigt werden und die Grundfunktionen IN-
SERT, UPDATE und DELETE aussehen können. 

Kapitel 7 liefert einen Überblick über Job Control. Sinniger Weise befassen wir uns hier nur mit den 
gängigen Parametern. Beschrieben werden die alltäglichen Operanden der JOB-, EXEC- und DD-
Anweisung. Darüber hinaus die nun schon seit längerem verfügbaren aber zu selten benutzten Spe-
cial-Statements wie IF-THEN-ELSE, private PROCLIB-Zuweisungen, INCLUDES, SETs und der-
gleichen mehr. 

Kapitel 8 enthält eine kurze Beschreibung der Datenstrukturen aller in den Demos benutzten Dateien 
für ein besseres Verständnis der Kodierbeispiele. 

Band-2, Syntax Referenz 

Hier wird die Syntax der diversen Befehle beschrieben. Anders, als im 2009 erschienenen REXXi-
kon, gibt es hier keine Trennung nach Zugehörigkeit mehr. Alle Befehle sind alphabetisch geordnet. 
Auf diese Weise lassen sich die Befehlsbeschreibungen sehr viel einfacher finden. Da es diverse 
gleichnamige Befehle aus unterschiedlichen Bereichen gibt, wurde die Beschreibung all dieser Teil-
bereiche im Referenzteil vereint. Für alle Befehle werden die Operanden über ein Syntaxdiagramm 
abgebildet. In der darüber liegenden Zeile wird seine Herkunft beschrieben (TSO-Kommando, 
REXX-Instruktion, ISPF-Befehl, JCL-Operand, etc.). Ich bitte um Verständnis, dass diese Beschrei-
bung nicht vollständig ist. Die Idee war, dass auch ältere Mitmenschen dieses Nachschlagewerk 
ohne fremde Hilfe transportieren können sollen. 

Literaturhinweis 

Ich möchte an dieser Stelle keine Hinweise auf Bestellnummern von Manuals der IBM geben. Häufig 
ändern sich die Titel und Buchnummern. Generell möchte ich auf den Server der IBM verweisen, 
auf dem Manuals zum Download zur Verfügung stehen. Viele Manuals stehen als Reference und 
Users Guide zur Verfügung. In der Reference wird im Allgemeinen die Syntax beschrieben. Im Users 
Guide stehen wertvolle Hinweise für die Anwendung. 

Und noch eine Bitte: Versuche nicht alles zu ergoogeln. Wer braucht 2,8 Millionen Treffer in 0,276 
Sekunden? Wer liest in den Treffern über die Seite 2 hinaus? Und wer kann richtig beurteilen, welche 
qualitativ fragwürdigen Einträge er vorgesetzt bekommt. Benutzt immer das Manual der IBM, das 
zur eingesetzten Version Ihres Betriebssystems passt. Ist das Manual zu alt, fehlen womöglich inte-
ressante Dinge. Ist es zu neu, wird das eine oder andere nicht funktionieren. Für gewöhnlich werden 



Inside z/OS Seite 8 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

alle IBM-Manuals im Unternehmen von der Systemtechnik auf einem Server für die Allgemeinheit 
zur Verfügung gestellt. Wenn nicht: Deine Systemprogrammierung hilft gerne weiter (oder Du such-
sat Dein Glück auf dem Book-Server der IBM, wenn Du weißt welche zOS-Version auf Deinem 
Rechner läuft). 

Verweise auf das Internet 

Die gängigsten NewsGroups. Hier treffen sich die absoluten Größen der Fachwelt zu den jeweiligen 
Themen. Sehr Empfehlenswert: 

Group Thema 

bit.listserv.ibm-main Generelle Mainframe-Themen 

bit.listserv.tsorexx Alles zu REXX unter TSO/E (und mehr) 

bit.listserv.ispf-l Alles zu ISPF 

bit.listserv.db2-l Alles zu DB2 

Interessante Homepages (alphabetisch): 

URL Inhalte / Betreiber 

www.cbttape.org MVS-Freeware 

www.cbttape.org/links.phtml Links von der Freeware-Seite 

www.mzelden.com/mvsutil.html Mark Zelden 

www.ibm.com/systems/z/os/zos/bkserv/ BookServer der IBM 

www.mvsforums.com Manuals und Links 

www.planetmvs.com David Alcock 

www.rexxla.org Rexx Language Association 

www.theamericanprogrammer.com Viele Links zu Literatur 

www.tsotimes.com TSO und ISPF 

 

  

http://www.rexxla.org/

1. TSO/E-REXX 

1.1. Allgemeines 

Mit REXX wird eine Sprache zur Verfügung gestellt, die in weiten Bereichen von z/OS Assembler 
ersetzt. War es früher undenkbar, in der Systemtechnik ohne fundierte Assemblerkenntnisse aus-
zukommen, ist es heute eher selten, dass die Mitarbeiter dort dieser Sprache mächtig sind. Sie wird 
auch nicht wirklich vermisst, da mit REXX sehr tief ins System eingegriffen werden kann.  

Bezieht man die Möglichkeiten von REXX in den Randbereichen wie System-Automation (REXX 
unter Netview) oder System-REXX mit ein, gibt es praktisch nichts, das man mit REXX nicht machen 
könnte. Und wer nicht hoch performante oder autorisiert laufende Programme braucht, wird Assemb-
ler nicht vermissen. 

1.1.1. Historisches 

REXX hatte seinen Ursprung im VM/CMS, einem Mainframe-Betriebssystem der IBM. Als Interpre-
ter-Sprache fand es dort in sehr kurzer Zeit viele Freunde. Entwickelt wurde REXX von Michael F. 
Cowlishaw (auch bekannt als Mike REXX). Die Sprache zeichnet sich durch einfache Strukturen und 
eine Vielzahl von eingebetteten Funktionen aus und ist leicht erlernbar. Viele Produkte aus dem 
Bereich Mainframe bieten ein API zu REXX. 

Ende der 80er Jahre des letzten Jahrhunderts fand REXX den Weg ins MVS und hat dort die Inter-
pretersprache CLIST verdrängt. CLISTen sind zwar auch heute noch in der TSO- und Netview-Um-
gebung anzutreffen, stammen aber meist aus der grauen Vorzeit. Neuentwicklungen werden in der 
Regel nicht mehr in CLIST geschrieben. 

Auf einen Nenner gebracht: REXX kann alles, was CLIST kann. REXX kann auch alles, was CLIST 
nicht kann. 

IBM lieferte sein PC-Betriebssystem OS/2 erstmals mit REXX anstelle von Basic aus. Auch zu DOS 
war REXX erhältlich. Unter UNIX/Linux wurden REXX-Derivate angeboten und heute noch kann mit 
Regina-REXX ein Interpreter für Linux und Windows kostenlos aus dem WEB geladen werden. 
REXX ist damit eine der wirklich wenigen Sprachen, die vollkommen plattformunabhängig ist, soweit 
man das überhaupt so sagen darf. Richtiger wäre wohl, dass REXX auf allen Plattformen verfügbar 
ist. Aber überall dort, wo nicht der Interpreter zuständig ist (wie beispielsweise für das Dateihand-
ling), gibt es natürlich gravierende Unterschiede zwischen den verschiedenen Plattformen. 

1.1.2. Drei kleine Wermutstropfen 

IBM hat den SAA-Standard (System Application Architecture - Konzept zur Standardisierung der von 
IBM entwickelten Software) gebildet. Befehle, welche dem SAA-Standard entsprechen, sind platt-
formunabhängig. Leider entsprechen nicht alle Funktionen in TSO/E-REXX dem SAA-Standard. 

Die zweite gewöhnungsbedürftige Situation ist die Dateiverarbeitung. Hier stehen in REXX unter 
TSO die Befehle ALLOC und EXECIO zur Verfügung. Auf anderen Plattformen wird an Stelle dieser 
Befehle die Dateiverarbeitung mit den Funktionen STREAM(), LINEIN(), LINEOUT() und LINES() 
durchgeführt. Generell können diese auch unter TSO benutzt werden, aber warum kompliziert, wenn 
es auch einfach geht. Generell möchte ich Ihre Aufmerksamkeit mehr auf den ALLOC-Befehl aus 
der Unix-Umgebung (USS) lenken. Erstens kann dieser in jedem Umfeld benutzt werden und zwei-
tens ist er deutlich vielseitiger als sein Verwandter aus dem TSO. 

Wer PIPEs aus Netview kennt, wird diese im TSO vergebens suchen. Es gibt dort die sogenannten 
Batch Pipes. Diese sind aber nicht direkt vergleichbar und zudem ein kostenpflichtiges Zusatzpro-
dukt, welches die meisten Unternehmen nicht gesondert anmieten. Generell können die PIPE-Funk-
tionen käuflich erworben werden, allerdings nicht über die IBM. Die meisten Unternehmen lehnen 
dies ab, da die Pipes unter TSO exotisch sind und keinen Standard darstellen. Verständlich. 

1.2. Synonyme 

Aktivitäten werden auf dem Mainframe über Synonyme durchgeführt. Anders als auf dem PC, wäre 
es hier unvorstellbar, die Namen von Dateien hart in ein Programm zu kodieren. Dieses müsste 
jedes Mal geändert werden, wenn sich der Dateiname ändert. 

Diese Systematik wird auch im Systemumfeld benutzt. Ladeaktivitäten für Programme, Panels, Ta-
bellen etc. werden über Synonyme. Wir sprechen in diesem Zusammenhang auch von den sog. DD-
Namen – eine Art Alias. Welche physischen Dateien sich dahinter verbergen, ist für eine Verarbei-
tung ohne Bedeutung. 

Oft stehen mehrere Dateien mit einem DD-Namen in Verbindung. In diesem Fall werden die Dateien 
in der Reihenfolge ihrer Verkettung gelesen. Klar, dass das nur für Lesen gilt. Für Schreiboperatio-
nen sollte sinniger Weise immer eine konkrete Datei definiert sein. 

Für eine funktionierende Umgebung, in welcher mit REXX und ISPF-Services gearbeitet werden 
soll, ist es unerlässlich, ein sauberes Umfeld aufzubauen. Dies geht in aller Regel bereits während 
der Logon-Phase vonstatten. Leider kann in Zuge des Logons nicht jede Anforderung für jede belie-
bige Applikation erfüllt werden. Oftmals muss die Applikation das von ihr benötigte Umfeld selbst 
aufbauen. Aus diesem Grund sollen hier zunächst eine Beschreibung der für eine Dialoganwendung 
benötigten DD-Namen und deren Funktionalität erfolgen. Diese Dateien haben PO-Format (partitio-
ned). Um Ärger zu vermeiden sollten die Dateien das Format Library (PDS/E) haben. Und wenn Du 
Dir selbst eine Freude machen willst, benutze PO/E-Type2 (erlaubt performantere Zugriffe auf große 
Directories und das Abspeichern von Generationen der Member (vergleichbar zu GDGs, muss man 
aber nicht). Dies macht aber nur Sinn, wenn Du über ein Tool für den Umgang mit Library Type-2 
verfügst. Ein sehr gutes gibt es auf www.CBTTAPE.org: PDSEGEN von Lionel Dyck. Ist wirklich 
hervorragend, kostet nichts und wird mit allen Quellcodes ausgeliefert. 

DDNAME Funktion 

SYSEXEC Aufruf von REXX-Programmen (Systemnah) 

SYSUEXEC Aufruf von REXX-Programmen (Userbezogen) 

SYSPROC Aufruf von CLISTen (Systemnah) 

SYSUPROC Aufruf von CLISTen (Userbezugen) 

ISPPLIB Aufruf von ISPF-Panels (Bildschirmmasken) 

ISPMLIB Aufruf von Nachrichten in ISPF-Dialogen (Messages) 

ISPPROF Der Variablenpool von ISPF 

ISPTLIB Zum Lesen von ISPF-Tabellen 

ISPTABL Zum Schreiben von ISPF-Tabellen. Sollte eindeutig sein 

ISPSLIB Zum Lesen von ISPF-Skeletons 

ISPFILE Zum Schreiben von File Tailoring Ergebnissen 

ISPLLIB Für den Aufruf von LOAD-Modulen in ISPF 

1.2.1. SYS[U]EXEC und SYS[U]PROC 

Der DD-Name SYS[U]EXEC wurde mit der Einführung von REXX unterstützt. Bis zu diesem Zeit-
punkt war CLIST die Interpretersprache im TSO. Um Alt und Neu besser voneinander trennen zu 

http://www.cbttape.org/

können, wurde von IBM empfohlen, CLISTs unter dem bisherigen DD-Namen SYS[U]PROC weiter-
hin zu rufen. REXX dagegen sollte über den DD-Namen SYSEXEC aufgerufen werden. 

Generell können natürlich auch REXXe aus SYSPROC sowie CLISTen gerufen werden. Aber an 
dieser Stelle eine persönliche Frage: Würdest Du Deine Sportschuhe nach dem Joggen ins Gemü-
sefach neben das Frischobst legen, nur weil es geht? Im Übrigen sind die Aufrufe von REXXen aus 
SYSEXEC performanter als aus SYSPROC. Es gibt aber durchaus Überlegungen, die zu REXX-
Aufrufen aus SYSPROC führen. 

In der Regel werden über SYSEXEC/SYSPROC Systembibliotheken des Unternehmens zugewie-
sen, benutzerspezifische REXX-/Clist-Bibliotheken dagegen über SYSUEXEC/SYSUPROC. 

1.2.2. ISPPLIB 

Ist in ISPF-Dialogen zuständig für das Laden von Bildschirmmasken. Es ist unerheblich, über wel-
chen Service die Masken geladen werden sollen. Es kann sich dabei um Datenpanels, Tabellenpa-
nels oder Auswahlpanels handeln. 

1.2.3. ISPMLIB 

Wird für den Aufruf von Nachrichten in einem ISPF-Dialog benutzt. Die Membernamen, in denen die 
Nachrichten gesucht werden, ergeben sich aus dem Alpha-Prefix der Nachricht und den ersten bei-
den numerischen Werten dahinter. Eine Nachricht AKUV001A beispielsweise wird innerhalb der 
ISPMLIB-Verkettung in einem Member AKUV00 gesucht. Mehr dazu bei der Beschreibung der 
Nachrichtenbehandlung. 

1.2.4. ISPTLIB und ISPTABL 

Diese DD-Namen sind für Tabellenverarbeitung vorgesehen.  

ISPTLIB wird zum Lesen von Tabellen benutzt. Dahinter können sich beliebig viele Dateien verber-
gen. Die Tabellen werden unter gleichem Namen als Members in ISPTLIB gesucht. 

ISPTABL wird für Schreiboperationen von Tabellen benutzt. Diese Zuweisung sollte eindeutig sein. 
Es macht nicht viel Sinn, für eine Schreiboperation mehrere Ausgaben zu definieren und dem TSO 
zu überlassen, irgendeine von diesen zu benutzen. 

1.2.5. ISPSLIB und ISPFILE 

Diese DD-Namen sind für File Tailoring (Skeleton-Verarbeitung) vorgesehen.  

ISPSLIB wird zum Lesen von Skeletons benutzt. Dateiverkettung ist möglich. Die Skeletons werden 
als namensgleiche Member in ISPSLIB gesucht. 

ISPFILE wird für Schreiboperationen im File Tailoring benutzt, wenn die Ergebnisse nicht temporär 
erstellt werden sollen. Diese Zuweisung sollte immer eindeutig sein.  

1.2.6. ISPPROF 

Hier wird der Applikationsspezifische ISPF-Profile-Variablenpool abgebildet. Der Membername bil-
det sich aus der Applikationskennung als Prefix und dem Suffix PROF. 

Der Variablenpool für ISPF-Standard ist ISPPROF.  

1.2.7. ISPILIB 

Die Image-Library. Von dieser haben viele ISPF-Programmierer noch nie gehört. Unter ISPILIB kön-
nen Images im GIF-Format hinterlegt werden die in graphischen Bereichen eines Panels abbildbar 
sind (benötigt eine Graphic Area im Panel und das GDDM muss installiert sein). 

Besonderheiten: 

Die DD-Namen ISPTLIB/ISPTABL und ISPSLIB/ISPFILE können umgangen werden, indem die ent-
sprechenden Befehle mit dem Operanden LIBRARY() kodiert werden und über davon abweichende 
DD-Namen Schreib-/Lesevorgänge durchführen. Dies verkompliziert aber meist eine Anwendung 
unnötig. 

1.2.8. Die Dateizuweisungstabelle 

Oftmals ist es eine große Hilfe, sich die Dateizuweisungstabelle anzuschauen. In einer Tabelle wer-
den DD-Namen und die dazugehörenden Dateinamen in der Reihenfolge ihrer Verkettung ange-
zeigt. In dieser Anzeige sind auch eine ganze Reihe praktischer Befehle möglich, wie beispielsweise 
das Suchen eines Members. 

1.2.8.1. DDLIST 

Wird benutzt, um die Zuweisungstabelle aus der Commandline eines Panels aufzurufen. DDLIST ist 
allerdings lediglich ein Eintrag in der System Command Table. Das eigentliche Programm dahinter 
ist das ISRDDN 

1.2.8.2. ISRDDN 

Programm für die Anzeige der Allocation Table. Da das DDLIST nur gerufen werden kann, wenn 
zum Aufrufzeitpunkt die entsprechende Commandtable durchsucht wird, steht es beispielsweise 
während des Trace eines REXX-Programms nicht zur Verfügung. Aber auch oder gerade hier, wäre 
die Info sehr hilfreich. Ist beispielsweise ein interaktiver Trace über "?R" aktiv, kann durch die Tas-
tatureingabe <address tso isrddn> die Anzeige der Allocationtable bewerkstelligt werden. 

1.3. Dateinamen 

Dateinamen sind Schall und Rauch, aber es gibt gewisse Empfehlungen, an die man sich halten 
sollte. Nicht, dass es nicht auch anders ginge, was ist schon schlimm daran, sprechende Namen zu 
benutzen. Am Ende würde es ja jeder verstehen… 

Bitte informiert euch über die Regeln für Dateinamen in eurem Unternehmen. Heute wird überall 
DFSMShsm eingesetzt. Unter Umständen verlierst Du Deine Dateien schneller, als Du sie erstellt 
hast. 

DSNAME Inhalt 
Userid().?.EXEC Aufruf von REXX-Programmen 
Userid().?.CLIST Aufruf von CLISTen 
Userid().?.PANELS ISPF-Panel-Bibliothek 
Userid().?.MSGS ISPF-Message-Bibliothek 
Userid().?.TABLES ISPF-Tabellen-Bibliothek 
Userid().?.SKELS ISPF-Skeletons 

 

 



Inside z/OS Seite 15 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

1.4. Programmaufrufe 

1.4.1. Explizite Programmaufrufe 

Für den expliziten Aufruf von REXXen steht das TSO-Kommando EXEC zur Verfügung. Damit kann 
eine REXX aus einer beliebigen Datei aufgerufen werden, einerlei, ob diese unter SYSEXEC/SYS-
PROC im System definiert ist, oder nicht. Einzige Bedingung: der Aufrufer muss die erforderlichen 
RACF-Rechte besitzen. 

Bei der Ausführung des Befehles werden die TSO-Richtlinien bezüglich Dateinamen berücksichtigt. 
Vollqualifizierte Dateien werden zwischen Apostrophe gesetzt. Teilqualifizierte Dateien werden ohne 
Apostrophe und ohne FLQ (First Level Qualifier) kodiert (siehe Auswirkungen des TSO-Befehles 
PROFILE). 

EXEC 'USER01.MVS.EXEC(KALENDER)' EXEC 

Auf diese Weise kann eine REXX aus jeder beliebigen Bibliothek gestartet werden.  

EXEC MVS(KALENDER) EXEC 

Ergänzt den Namen der Bibliothek nach folgenden Standards: Als First Qualifier wird die Prefix-
Angabe des TSO-Profils benutzt. Der 2. Qualifier wird übernommen. Als dritter Qualifier wird EXEC 
eingesetzt, da es sich aufgrund des Parameters EXEC um eine REXX handelt und EXEC der Stan-
dard-Low Level Qualifier LLQ) ist.  

1.4.2. Implizierte Aufrufe 

Sie sind möglich, sofern die Bibliothek, aus welcher die REXX gerufen werden soll, unter 
SYS[U]EXEC zugewiesen ist (wir wollen uns für den Rest des Buches davon trennen, dass die Auf-
rufe auch über SYS[U]PROC möglich sind, es sei denn, Du legst Deine Joggingschuhe… wir lassen 
das lieber). Dementsprechend kann der Aufruf des Programms unter TSO durch Angabe des Pro-
grammnamens erfolgen: 

KALENDER<enter> 

Dabei kommt es zu folgender Situation: Der TSO Befehlsprozessor kann in dieser Form Aktivitäten 
aus drei verschiedenen Bereichen erzeugen und hält dabei eine bestimmte Suchreihenfolge ein: 

1. KALENDER wird im Bereich der TSO-Kommandos gesucht. Wird ein gleichnamiges gefun-
den, wird es ausgeführt. 

2. Gibt es kein gleichnamiges TSO-Kommando, wird die Linkliste (Standard-Lademodulbiblio-
theken im MVS) nach einem gleichnamigen Modul durchsucht. Dies ist in der Regel ein grö-
ßerer Aufwand, da die Suche über eine Vielzahl von Bibliotheken erfolgt, in denen jeweils 
mehrere Tausend Member vorhanden sein können. Gibt es eine namentliche Übereinstim-
mung, wird intern der TSO-Befehl CALL ausgeführt und das Lademodul wird aufgerufen. 

3. Ist die Suche innerhalb der Linkliste erfolglos, wird die Verkettung von SYSPROC/SYSEXEC 
durchsucht. Erfolgt die Zuweisung mit Verstand (via EXECUTIL) wird die SYSEXEC-Verket-
tung vor SYSPROC durchsucht. 

4. Wird auch hier nichts gefunden, wird eine Fehlermeldung ausgegeben. 

An dieser Stelle kann jeder ein wenig zu mehr Leistung zum Nulltarif beitragen. Ist bekannt, dass 
eine REXX aufgerufen werden soll, kann dies mit einem führenden %-Zeichen kenntlich gemacht 

werden. In diesem Fall wird nicht nach TSO-Befehlen oder Programmen gesucht. Das hat auch den 
Vorteil, dass REXXe namensgleich zu TSO-Kommandos benannt werden können. Erfolgt der Aufruf 
ohne %-Zeichen, wird der TSO-Befehl ausgeführt. Erfolgt er mit, wird die REXX aufgerufen. 

%KALENDER<enter> 

Weiterer Vorteil beim Aufruf mit % ist, dass bei Programmstart intern ein Clear Screen Kommando 
abgesetzt wird. 

1.5. Merkmale von REXX 

REXX zeichnet sich durch einige nicht selbstverständliche Merkmale aus: 

REXX-Programme sind leicht zu lesen, die Sprache ist schnell erlernbar. Die meisten Befehle sind 
simple Worte aus dem Englischen, keine Abkürzungen, keine Hieroglyphen (SAY, PULL, IF-THEN-
ELSE, DO, END, EXIT). 

Bei der Kodierung gibt es kaum formale Zwänge. Die Anweisungen können in Groß- oder Kleinbuch-
staben oder gemixt geschrieben werden. Zeilen können nach Belieben eingerückt kodiert werden. 
Weder für den Beginn einer Instruktion, noch für die Spalte einer Fortsetzungszeile gibt es vorge-
schriebene Spaltenpositionen. Auch innerhalb einer Zeile können Spaltenbereiche durch Kommen-
tardefinitionen übergangen werden. Generell könnten sogar mehrere Befehle in einer Zeile kodiert 
werden. Von dieser Möglichkeit ist in der Praxis wegen der schlechten Lesbarkeit abzuraten (Aus-
nahmen sind denkbar, wo die Lesbarkeit verbessert wird). 

Eine Instruktion kann, sofern sie in einer Zeile keinen Platz findet, über mehrere Zeilen fortgesetzt 
werden. Als Ankündigung der Fortsetzung genügt es, eine Zeile mit Komma(,) enden zu lassen. Die 
Fortsetzung kann in der nächsten Zeile an beliebiger Position beginnen. 

Ob REXX-Instruktionen in Sätzen fester oder variabler Länge hinterlegt sind, spielt für den Interpre-
ter keine Rolle. Üblich sind in den meisten Unternehmen aber partitioned Datasets (FB80). Gönn Dir 
den Luxus und benutze Library Type-2. Und wenn Du ein geeignetes Tool für die Nutzung der Ge-
nerationsführung unter z/OS habst (PDSEGEN von Lionel Dyck gibt es kostenlos auf 
CBTTAPE.ORG) kann auch das durchaus sinnvoll sein. Unter zOS 3.1 soll auf die Generationen 
auch via PDF sinnvoll zugegriffen werden können. 

1.5.1. Variablen 

Variablen müssen nicht vordefiniert werden und können während des Programmverlaufes ihre in-
haltliche Länge, und ihren Datentyp (character oder numeric) ändern. Genau genommen gibt es in 
REXX gar keine numerischen Felder, sondern nur Strings. Ein Abbruch wegen unbekannter Variab-
len in REXX ist nicht denkbar, es sei denn, dies ist über eine entsprechende NOVALUE-Condition 
so gewollt. Eine nicht belegte Variable in einem REXX-Programm zeigt inhaltlich immer auf ihren 
zur Großschreibung konvertierten Eigennamen. 

1.5.2. Testhilfen 

Betrachtet man REXX, drängt sich immer wieder der direkte Vergleich zu CLIST auf. In Bezug auf 
Testhilfen bietet REXX hier weit mehr und komfortablere Verfahren, Fehler im Ablauf zu erkennen. 
Eine Möglichkeit bietet das TRACE-Kommando, das mit seinen vielfältigen Optionen für jede Situa-
tion die richtige Unterstützung bietet 

1.5.3. Eingebettete Funktionen 

REXX bietet eine Vielzahl von Funktionen. Gerade in den Anfängen gewinnt man den Eindruck, für 
jedes Problem die Lösung durch die Funktionsvielfalt bereits fertig präsentiert zu bekommen. Tritt 
nach der ersten Euphorie die Phase der Ernüchterung ein, stellt man fest, dass die eine oder andere 
Funktion mehr durchaus wünschenswert gewesen wäre (typisch Anwender: gibt es sehr wenig Kom-
fort, gibt man sich damit zufrieden. Ist eine Sprache sehr komfortabel, verlangt man das Unmögliche 
von ihr). Kein Grund zur Frustration. REXX erlaubt auf eine sehr simple Weise, den Standard-Funk-
tionsumfang durch eigen geschriebene Funktionen nach Belieben zu erweitern (UDF’s - User Defi-
ned Functions). Beachtet man dabei einige Regeln, kann so ein Eigenprodukt wahlfrei als Unterpro-
gramm, Funktion oder als Kommando benutzt werden. Mehr dazu im Abschnitt »Modularisierung«. 

1.5.4. Arithmetik 

REXX erlaubt alle gängigen Formen arithmetischer Operationen. Dabei werden Divisionen mit De-
zimalergebnissen ebenso unterstützt wie integer Ergebnisse, Potenzrechnungen und Exponential-
schreibweisen (siehe unter Arithmetische Operatoren). 

1.5.5. Zeichenketten-Manipulationen 

Unter den vielfältigen Funktionen werden viele komfortable Möglichkeiten zur Zeichenmanipulation 
angeboten. Hier hat man sogar häufig den Eindruck, dass des Guten zu viel getan wurde. Oftmals 
können, um zum gleichen Ergebnis zu kommen, unterschiedliche Funktionen eingesetzt werden, die 
sich oft nur in der Reihenfolge der Parameterangaben unterscheiden. Der Grund hierfür ist vermut-
lich nicht darin zu suchen, dass die geistige Flexibilität der Anwender unter Beweis gestellt werden 
sollte, sondern liegt wohl eher an einem Mangel an Kommunikation zwischen den Entwicklern bei 
der Programmierung des Interpreters. 

1.6. Befehlsausführung unter einem Interpreter 

Dies hat Vor- und Nachteile. Einerseits ist das Schreiben von Programmen, das Austesten und Än-
dern sehr simpel. Die einzelnen Befehle werden zum Laufzeitpunkt vom Interpreter in Maschinen-
code umgewandelt. So entfällt der bei Sprachen wie Assembler, Cobol, PL/I oder Fortran nötige 
Übersetzungs- und Bindelauf (Compile und Link). Die Programme können getestet werden, sobald 
der erste Befehl kodiert ist.  

Andererseits bringt die Ausführung unter einem Interpreter Nachteile in punkto Performance mit sich. 
Nicht zu übersehen ist auch, dass für die Ausführung immer lesbarer Quellcode verfügbar sein muss, 
was unter den TSO-Hackern oft ungeahnte kriminelle Energien freisetzt. Beides lässt sich in der 
Praxis befriedigend lösen, da von IBM auch für REXX ein Compiler angeboten wird. Kompilierte 
REXX-Programme sind schneller im Ablauf (wenn natürlich nicht so schnell, wie übersetzte Cobol- 
oder Assembleranwendungen). Für viele Programme spielt dies eine untergeordnete Rolle. Im Zwei-
felsfall kann eine REXX-Anwendung auch in einem Batch-Adressraum zur Ausführung kommen, 
sofern einige Dinge beachtet werden (siehe REXX im Batch). Viel wichtiger ist dagegen, dass REXX-
Programme nicht im Quellcode zur Verfügung gestellt werden müssen, und so für den Anwender 
weder les- oder änderbar sind.  

Auch, wenn REXX-Programme generell unter dem Interpreter laufen sollen, empfiehl es sich, zur 
Syntaxprüfung den Compiler zu benutzen, da dieser auch Fehler erkennt, welche dem Interpreter 
verborgen bleiben und unter Umständen erst in bestimmten Konstellationen zur Laufzeit auftreten. 

In einem REXX-Programm können unterschiedliche Sprachkomponenten vorkommen: 

▪ REXX-Instruktionen 

▪ Wertzuweisungen 

▪ eingebettete Funktionen 

▪ benutzerdefinierte Funktionen 

▪ Sprungziele (Label oder Marken) 

▪ STACK-Funktionen, wie die Aufnahme von Daten aus I/O-Operationen  

▪ Befehle und Funktionen aus dem Umfeld (System Environment) von REXX. 

Dieses Umfeld ist im allgemeinen TSO. Generell kann aber auch Netview,der Editor (ISREDIT), 
ISPF, SDSF oder DB2 ankodiert werden.  

1.7. Literale 

Literale sind Zeichenketten, die von Apostrophen(') oder Quotes(") eingegrenzt werden. Grundsätz-
lich wäre eine der beiden Möglichkeiten ausreichend, da die Zeichen durch doppeltes Tippen zum 
Textbestandteil innerhalb des Literals werden. Die Optik sieht aber auch für den geübten Leser ei-
genartig aus. Die Anforderungen, Apostrophe(') innerhalb der Zeichenketten als Textbestandteil zu 
benutzen, treten oft auf, da TSO-Kommandos innerhalb eines REXX-Programms im Regelfall als 
Literale hinterlegt sind, damit sie nicht vom REXX-Interpreter aufgelöst werden. TSO-Kommandos 
enthalten Apostrophe('), wenn innerhalb der Befehle vollqualifizierte Dateinamen anzugeben sind. 
Werden diese Namen über Variable geliefert, wird das Chaos der Schreibweise perfekt und nur 
durch die richtige Anwendung der Literal-Begrenzer gemildert. Wird ein Literal mit einem Begren-
zungszeichen begonnen, wird alles, was folgt, als Teil des Literals interpretiert, bis das zur Einleitung 
der Zeichenkette benutzte Begrenzungszeichen ein zweites Mal kodiert wird. 

"Zeichenkette" 

'Zeichenkette aus mehreren Token gebildet' 

"Das wäre ebenso falsch' 

'wie das" 

"Heinzi's kleine Lotterie und" 

'.....sagte:"BLABLA" sind dagegen OK.' 

Im folgenden Beispiel soll eine Dateizuweisung durchgeführt werden, bei welcher der Name der 
Datei durch eine Variable geliefert wird. Entscheide selbst, welcher der beiden nachfolgenden Vari-
anten für die Schreibweise eines TSO-Kommandos Du den Vorzug geben würdest. Richtig sind 
beide: 

"ALLOC DD(TEMP) DA('"datei"') SHR REUSE" 

'ALLOC DD(TEMP) DA(''datei'') SHR REUSE' 

In REXX können Literale auch als HEX-Strings gekennzeichnet werden. In diesem Fall dürfen nur 
die Buchstaben A bis F, sowie die Ziffern 0 bis 9 im Literal vorkommen. Ob die Buchstaben groß 
oder klein geschrieben sind, spielt dabei keine Rolle. In der Zeichenkette enthaltene BLANKs, sofern 
sie an der Bytegrenze stehen, verbessern die Lesbarkeit, werden vom Befehlsprozessor aber igno-
riert. Zur HEX-Zeichenfolge wird das Ganze durch ein unmittelbar hinter dem Literal-Begrenzungs-
zeichen stehendes großes oder kleines X. Beispiele für HEX-Zeichenfolgen sind: 

'c1b4'x  oder  'C0F'x  oder  "C 03 FF"X 

1.8. Worte (Token) 

WORTE oder Token im Sinne von REXX sind Zeichenketten beliebiger Länge, wobei es natürlich 
theoretische Grenzen gibt, die in der Praxis eine untergeordnete Rolle spielen. Begrenzt werden 
Worte durch Komma(,), BLANK( ) oder durch sich selbst. Inhaltlich können Worte aus Zahlen, Buch-
staben oder Sonderzeichen (auch gemischt) zusammengesetzt sein. Alle folgenden Beispiele be-
stehen aus jeweils fünf Worten: 

"Der Pferd hat vier Bein" 

summe = 5 + 7 

"A entspricht dem EBCDI-Code 'C1'x" 

1.9. Variablen 

Variable sind Platzhalter. Sie stehen für Werte, die erst zur Laufzeit gebildet werden. Eine Variable 
wird als solche erkannt, weil sie außerhalb von Literal-Begrenzungszeichen (' oder") steht. 

Die Namen von Variablen sind praktisch unbegrenzt (250 Byte). Die inhaltliche Länge ist nicht be-
grenzt, sofern sie im virtuellen Adressraum der TSO-Sitzung Platz finden. 

Zur Namensvergabe dürfen Buchstaben (nicht Case-sensitiv), Ziffern und die Sonderzeichen Un-
derscore(_), Hash(#), Dollar($), At(@), Punkt(.) und Questionmark(?) genutzt werden. Die erste 
Stelle des Namens darf keine Zahl und kein Punkt sein. Wie groß auch immer der Anreiz ist, Vari-
ablennamen unter REXX sehr lang mit Einbeziehung der Sonderzeichen zu schreiben, -oft ergibt 
sich dadurch eine deutlich verbesserte Lesbarkeit-, so ekelhaft kann dies werden, wenn man sich 
im Nachhinein entschließt, ISPF-Services zu implementieren. Dort sind die Namen von Variablen 
auf acht Byte begrenzt. Außerdem dürfen sie nur aus Buchstaben, Zahlen und den nationalen Zei-
chen gebildet werden, wobei das erste Byte ein Buchstabe sein muss. Berücksichtigt man diese 
Restriktionen bereits von Anfang an, sind Programmerweiterungen zu einem späteren Zeitpunkt un-
problematisch. Gibt es bei der Namensvergabe von Variablen zwischen REXX und ISPF Unverträg-
lichkeiten, kann die entsprechende Komponente mit REXX nicht unmittelbar über die Variablen kom-
munizieren. 

Wie immer Sie sich bei der Vergabe von Variablennamen entscheiden, benutzen Sie sprechende 
Namen. Aus dem Namen der Variablen sollte auch für einen Außenstehenden abzuleiten sein, wozu 
sie genutzt werden soll.  

1.9.1. Empfehlung für Variablennamen 

Da es in Unternehmen oft keine starren Regeln für die Benennung von Variablen gibt, oder sich die 
Mitarbeiter nicht daran halten, hier einige Angebote zur Namensvergabe: 

Sxxxxxxx für Variable, die eine Schalterfunktion haben (der Inhalt variiert nur zwischen 0 und 1, 
oder J und N) 

Cxxxxxxx für Variable, die eine Zählfunktion übernehmen (C wie COUNT; natürlich könnte auch 
Z wie ZÄHLER genutzt werden, entscheide Dich aber konsequent für eine Lösung). 

_xxxxxxx für temporäre Variable, die nur kurzfristig als Träger eines Zwischenergebnisses ge-
nutzt werden und zu einem späteren Zeitpunkt im Programm bedenkenlos über-
schrieben werden können. 

Vxxxxxxx für symbolische Programmvariable innerhalb REXX. Je nach Bedarf und Einbindung 
von ISPF könnte man den Ursprung eines Wertes mit 

Pxxxxxxx einem ISPF-Panel oder durch 

Txxxxxxx einer Dialog Manager Tabelle zuweisen. 

Die restlichen sieben Byte sollten den Inhalt der Variablen beschreiben.  

1.9.2. Spezielle Variablen 

Der Interpreter füllt situationsabhängig drei spezielle Variablen mit Werten, die im Programm abge-
fragt werden können. Deren Namen und Bedeutung sind:  

Variable Bedeutung 

RC Return Code der letzten ausgeführten Instruktion 

RESULT Aus einem Unterprogramm durch RETURN rückgereichter Wert 

SIGL Quellencode-Zeilennummer, die aufgrund einer Sprungbedingung verlassen 
wurde. Keine dieser Variablen hat einen Grundwert. 
In der Regel wird die Sprungbedingung durch ein SIGNAL ON oder CALL ON 
veranlasst  (siehe SIGNAL, "Test und Fehlerbehandlung"). Es wäre auch ein 
Function-Call denkbar. Auf diese Form wird aber nicht weiter eingegangen, da 
dies nicht der regulären Nutzung entspricht. 

1.9.3. Wertzuweisungen 

Variablen können in REXX auf unterschiedliche Weisen gefüllt werden. Die häufigste Art ist die di-
rekte Zuweisung. Wertzuweisungen für Variable können aber auch das Ergebnis eines PARSE-Be-
fehles, einer Dateiverarbeitung (Lesen) oder eines Funktionsaufrufs sein. Sehen wir uns einige For-
men der Variablenzuweisung an: 

Zuweisung Variableninhalt 
A = 5 5 

B = 3.7 3,7 (Dezimal) 

C = 3E6 3000000 (= 3 mal 106 

D = A + 7 12 

E = D + 3 * 2 18 

F = (D + 3) * 2 30 

G = name NAME 

H = "Name" Name 

I = "Wun" Wun 

J = "der" der 

K = 'bar' bar 

L = I"der"K Wunderbar 

M = I J K Wun der bar 

N = IJK IJK 

An dieser Stelle treffen wir auf ein: Werden die Variablen I,J und K durch BLANK getrennt aneinan-
dergereiht, werden sie als jeweils eigene Variable erkannt. Das BLANK zwischen den Variablen wird 
übernommen. Wird auf die trennenden BLANKs verzichtet, kann REXX nicht mehr drei unabhängige 
Variablen erkennen, sondern unterstellt, das IJK eine neue, nicht gefüllte Variable ist, die somit auf 
ihren zu Großbuchstaben konvertierten Eigennamen zeigt. Um die Inhalte zweier Variablen unmit-
telbar zu verbinden, wird als Verkettungszeichen zweifaches Ausrufezeichen(!!) oder NULLSTRING 
kodiert. Woraus folgt: 

Zuweisung Variableninhalt 
O = I !! J!!K Wunderbar 

P = I""J""K Wunderbar 

Q = I''J''K Wunderbar 

R = 2 b7"x 02B7(hex) 

 

1.9.4. Rücksetzen von Variablen 

Soll eine Variable auf ihren Urzustand zurückgesetzt werden, ist dies nicht durch var="" möglich. 

In diesem Fall wäre der Inhalt der Variablen NULLSTRING. Ausgangssituation war aber: Die Vari-
able zeigt auf ihren groß geschriebenen Eigennamen. Diese Situation kann über den Befehl DROP 

var herbeigeführt werden. 

1.9.5. Compound-Variablen (Stems) 

Compound-Variablen bilden eine Besonderheit in REXX. Ihre Namen sind mehrgliedrig. Die einzel-
nen Teile werden durch Punkt(.) getrennt. Aus wie vielen Namensteilen eine Compound-Variable 
besteht, spielt keine Rolle. Die einzige Einschränkung besteht in der Gesamtlänge des Namens, der 
250 Byte nicht überschreiten darf. Über die Compound-Variablen können, zumindest logisch, Spei-
chertabellen nachvollzogen werden. Am besten sehen wir uns das nachfolgende Beispiel einer Ta-
bellendemo an: 

Tabelle TEXT 

1 2 3 4 5 

Hallo Servus Bye, bye Salü Ciao 
 

/* REXX * TABDEMO  

*/ 

TEXT.1="Hallo" 

TEXT.2="Servus" 

TEXT.3="Bye, bye" 

TEXT.4="Salü" 

TEXT.5="Ciao" 

Pos = random(1,5) 

say TEXT.pos 

exit 

Soll ein Stem als Gesamtes gefüllt oder zurückgesetzt werden, kann dies über ihren gemeinsamen 
Namensteil geschehen. Um alle denkbaren Variablen eines Stammes PLZ mit einer 0 vorzubelegen, 
genügt die Zuweisung PLZ.=0. Jede denkbare Compound-Variable, sofern sie mit PLZ. beginnt 

(PLZ.A.B.TRALLALA.IRGENDWAS, etc.) enthält den Wert 0. Ebenso lassen sich sämtliche Variab-
len eines gemeinsamen Stammes durch DROP PLZ. wieder auf Grundstellung bringen. 

Ein Beispiel für eine mehrdimensionale Tabelle, bei der eine Wortkette als Index benutzt wird, wel-
che auf mehrgliedrige Stems zeigt, kann im Abschnitt "Dateiverarbeitung" betrachtet werden. 

1.10. Operanden 

Operanden werden in drei Gruppen eingeteilt: 

▪ Arithmetische Operanden 
▪ Vergleichsoperanden 
▪ Boolesche Operanden 

1.10.1. Arithmetische Operanden 

Sie werden zur Durchführung von Rechenbefehlen genutzt. In größeren arithmetischen Ausdrücken 
erfolgt die Auflösung des Ausdruckes gemäß der Gewichtung der Operanden. Die nachfolgende 
Aufstellung zeigt die möglichen arithmetischen Operatoren und ihre Bedeutung. Die Reihenfolge von 
oben nach unten entspricht der Gewichtung. In dieser Reihenfolge erfolgt die Auflösung des Aus-
druckes. Soll davon abgewichen werden, können um Teile des Ausdruckes runde Klammern gesetzt 
werden. Klammern werden von innen nach außen gelöst. 

Operator Bedeutung Ausdruck Ergebnis 

** Potenzierung 2**4 16 

// Divisionsrest 29 // 4 1 

* Multiplikation 3 * 7 21 

% Division integer (ganzzahlig) 23 % 5 4 

/ Division (Dezimalbruch) 10 / 4 2.5 

+ Addition 4 + 7 11 

- Subtraktion 13 – 5 8 

 

Der Ausdruck 49-2**5*3/2 liefert 1 weil 49-(((2**5)*3)/2) 

Um hier auch für das Leben lernen zu können, möchte ich einen Ausspruch eines Bekannten nicht 
vorenthalten, der meinem Sohn immer sagte: "Die wichtigen Rechenarten sind PLUS und die MUL-
TIPLIKATION, sie mehren den Inhalt des Geldbeutels. Auf MINUS und die DIVISION kannst Du 
ruhig verzichten, denn nehmen lassen wir uns nichts und teilen tun wir auch nicht gerne". 

Ein Mehr an Arithmetik ist in der Regel nicht nötig. Wer etwas vermisst, sollte sich hinterfragen, ob 
er mit REXX das richtige Werkzeug gewählt hat (frei nach dem Motto: wenn man nur einen Hammer 
hat, sieht alles wie ein Nagel aus). 

Wer aber meint, er müsse die Umlaufbahnen von Planeten in REXX berechnen, wird im Internet 
auch Sinus-/Cosinus-/Wurzel- und sonstige Rechenfunktionen finden. Viel Erfolg. Im Übrigen gibt 
es auch diverse Funktionsbibliotheken, über welche man REGEXe und vieles mehr benutzen kann. 

1.10.2. Vergleichsoperanden 

Sie werden benutzt, um Ausdrücke zu vergleichen. Anders als in vielen anderen Sprachen können 
in REXX keine Buchstabenkürzel als Vergleichsoperanden genutzt werden (EQ für =). Dies ist ge-
rade für CLIST-Umsteiger sehr lästig, aber man gewöhnt sich ja an alles. Die möglichen Ver-
gleichsoperanden und ihre Bedeutung können aus der nachfolgenden Tabelle entnommen werden: 

Aussage Operator 

Gleich = 

Ungleich ^= 

Identisch == 

Nicht identisch ^== 

Kleiner als < 

Strikt kleiner << 

Kleiner oder gleich <= 

Nicht kleiner ^< 

Strikt nicht kleiner ^<< 

Größer > 

Strikt größer >> 

Größer oder gleich >= 

Strikt größer oder gleich >>= 

Nicht größer ^= 

Strikt nicht größer ^>> 

 

Offenbar trifft auch in REXX zu: Alle sind gleich, manche sind gleicher! Der Unterschied zwischen 
GLEICH und IDENTISCH liegt bei Character Strings in führenden BLANKs, bei numerischen Werten 
in führenden Nullen. Bei Prüfung auf GLEICH werden sie ignoriert. Bei IDENTISCH muss eine voll-
ständige Übereinstimmung der Zeichen erkannt werden. Gleiches gilt auch für Vergleiche wie 
STRIKT KLEINER. 

Vergleichsoperatoren werden in Befehlen wie IF, WHEN, DO WHILE und DO UNTIL zur Prüfung 

einer Bedingung oder zur Steuerung einer Schleife eingesetzt. Da die Vergleiche nach dem EBCDI-
Code durchgeführt werden, können Äpfel mit Birnen verglichen werden (Character und numerische 
Werte). Dabei ist aber zu beachten:  

ANTON  < BERTA 

berta  < Anton 

hugo  < Hugo 

Oskar  < 1 

Robert < Roberta 

Oder generell: 

Sonderzeichen < LowerCase < UpperCasse < Zahlen 

1.10.3. Boolesche Operanden 

Sie bilden die dritte Gruppe und werden benutzt, um mehrere Ausdrücke miteinander zu verknüpfen. 
Mögliche Angaben sind: 

Operator Bedeutung 

& UND-Verknüpfung 

! ODER-Verknüpfung 

&& EXCLUSIVE-ODER-Verknüpfung 

^ als Prefix Negation 

 

Durch diese Operatoren können die Ergebnisse von Vergleichen (TRUE(1) oder FALSE(0)) mitei-
nander verbunden werden, und man erhält wiederum ein RICHTIG(1) oder FALSCH(0). Aus der 
nachfolgenden Tabelle kann die Funktionsweise der Verknüpfungen abgelesen werden: 

 UND (&) ODER(!) EX-ODER(&&) Negation 

 Richtig Falsch Richtig Falsch Richtig Falsch  

Richtig 1 0 1 1 0 1 0 

Falsch 0 0 1 0 1 0 1 

 

Anders ausgedrückt: Eine UND-Verknüpfung ist wahr, wenn beide Bedingungen zutreffend sind. 
Eine ODER-Verknüpfung wird bejaht, wenn zumindest eine Bedingung erfüllt ist. Ein TRUE für die 
EXKLUSIV-ODER-Verknüpfung erhält man, wenn ausschließlich eine Bedingung erfüllt ist. 

Wendet man die Booleschen Regeln konsequent im täglichen Leben an, wäre interessant zu erfah-
ren, wie die deutsche Rechtsprechung hierüber denkt, denn Schilder wie Betteln und Hausieren 
verboten, oder Zelten und Feuermachen verboten, würden grundsätzlich das Betteln erlauben, so-
fern man nicht gleichzeitig hausiert, oder es wäre nichts gegen das Zelten einzuwenden, sofern man 
dabei kein Feuer macht (und umgekehrt). Im Sinne des Herrn Boole müssten diese Vorschriften 
lauten: Betteln oder Hausieren, sowie: Zelten oder Feuermachen verboten. Würden die Bedingun-
gen dagegen mit einem exklusiven ODER verbunden werden, wäre sowohl das Betteln und Hausie-
ren, als auch das Zelten und Feuermachen erlaubt, solange man immer beides in Verbindung und 
nicht getrennt voneinander durchführt. 

1.11. Kommentare 

Kommentare werden vom Befehlsprozessor ignoriert und müssen immer zwischen /* und */ einge-
schlossen werden. Es spielt dabei keine Rolle, ob als Kommentar eine Zeile oder ein Teil davon 
definiert wird, oder ob sich der Kommentar über mehrere Zeilen erstreckt. 

1.12. Befehlsbegrenzer 

Das Zeichen Semikolon(;) wird in REXX als Befehlsbegrenzer eingesetzt. Wenn wir davon ausge-
hen, dass in einer Zeile nicht mehr als ein Befehl steht, kann bei der Kodierung auf das Zeichen 
verzichtet werden. Wird es kodiert, ist es nicht störend. Werden dagegen mehrere Befehle in einer 
Zeile kodiert, müssen sie zwingend mit Semikolon voneinander getrennt werden. Die Lesbarkeit 
eines Programmes verschlechtert sich durch solche Unarten ganz radikal. 

1.13. Instruktionen 

Instruktionen sind Anweisungen an den Interpreter, die ein bestimmtes Ergebnis erzeugen. Im End-
effekt sind Instruktionen kleine Programme, die ablaufen. Das Ergebnis einer Instruktion kann im 
Programm frühestens in der nächsten Zeile ausgewertet werden. Formal werden Instruktionen in 
Form von Schlüsselworten geschrieben. Abhängig von der Instruktion und dem gewünschten Er-
gebnis, können Parameter an die Instruktion angefügt werden. 

SAY "Hier steht beliebiger Text" 

Generell unterscheiden wir zwischen 

▪ Instruktionen, die der REXX-Sprachinterpreter ausführt 

▪ Instruktionen, die zur Ausführung an das TSO weitergeleitet werden (extended TSO-Func-
tions) 

▪ TSO-Kommandos, beziehungsweise Anweisungen an eine andere eingestellte oder adres-
sierte Umgebung 

Die beiden letzten werden als Literale kodiert. Enthalten diese Instruktionen Variablen, sind die Quo-
tes vor der Variablen zu beenden und nach der Variablen fortzuführen, da anderenfalls die Variable 
nicht als solche erkannt werden kann. 

1.14. Funktionen 

Funktionen sind letztlich nichts wesentlich anderes als Instruktionen. Auch bei einem Funktions-
aufruf wird im Endeffekt ein Programm ausgeführt. Funktionen haben aber eine grundsätzlich an-
dere Schreibweise und können immer nur Teil einer Instruktion sein. Eine Funktion außerhalb einer 
Instruktion wird vom Interpreter mit einer Fehlermeldung und der Ausgabe eines Return Codes(-3) 
quittiert (Ein Return Code -3 wird gemeldet, wenn der Interpreter eine Instruktion oder Funktion nicht 
ausführen kann, sie an die aktive Systemumgebung weitergibt, und die Aktion auch dort unbekannt 
ist). 

An den Namen der aufzurufenden Funktion werden immer eine öffnende und schließende Klammer 
angehängt. Abhängig von der Funktion und den speziellen Bedürfnissen, können innerhalb der 
Klammer eine oder mehrere Angaben erfolgen. Durch die Angaben in der Klammer kommuniziert 
der Aufrufer der Funktion mit der Funktion (Datenübergabe). 

Ein weiterer Unterschied besteht in der Ablaufsteuerung des Interpreters bei Funktionsausführun-
gen. Wird nach ausgeführter Instruktion die nächste Zeile des Quellencodes verarbeitet, sieht es bei 
Funktionen anders aus. Liest der Interpreter in einer Instruktionszeile eine Funktion, ruft er sie auf 
und bringt das Ergebnis der Funktion (den von der Funktion zurückgelieferten Wert) innerhalb der 
Instruktionszeile an die Stelle der Funktion. Im weiteren Sinn könnten wir dieses Vorgehen mit der 
Substitution von Variablen vergleichen. 

SAY DATE()  liefert tt mmm jjjj (24 Mar 2009) 

SAY DATE('E') liefert tt/mm/jj  (24/03/04) 

Diese Unterschiede kann man sich zunutze machen, wenn man selbst Funktionen oder Unterpro-
gramme schreibt. Berücksichtigt man bei der Kodierung einige grundsätzliche Dinge, kann einzig 
durch die Art des Aufrufes entschieden werden, ob eine Funktion oder ein Unterprogramm gestartet 
werden soll. Mehr dazu im Abschnitt Modularisierung. 

Eine Funktion wird vom Interpreter daran erkannt, dass dem Namen der Funktion immer unmittelbar 
ein Klammernpaar folgt. Sollen Informationen an die Funktion übergeben werden, sind diese inner-
halb der Klammern anzugeben.  

Eine Funktion, die alleine in einer Zeile steht (außerhalb einer Instruktion) wird vom REXX-Interpreter 
mit einem RC-3 zurückgewiesen. Zuweilen tut dies der Wirksamkeit keinen Abbruch (extended TSO-
Functions). Der besseren Lesbarkeit willen sollte auf diese Schreibweise aber generell verzichtet 
werden. 

REXX kennt über 50 sogenannter Built-in-Funktionen, die sich in unterschiedliche Kategorien ein-
teilen lassen: 

• Arithmetische Funktionen 

• Vergleichsfunktionen 

• Funktionen zum Konvertieren von Daten 

• Formatierungsfunktionen 

• Funktionen zur String-Manipulation 

• Nicht zuordenbare restliche Funktionen 

Darüber hinaus sind die erweiterten TSO-Funktionen verfügbar. 

1.15. Systemumgebungen (Environments) 

REXX kann unter verschiedenen Umgebungen laufen. Abhängig vom jeweiligen Umfeld können un-
ter REXX die Dienstleistungsangebote der Umgebung genutzt werden. Generell wären möglich 
MVS, TSO, ISPEXEC, ISREDIT, LINK, ATTACH und NETVIEW. 

Abhängig davon, in welchem Adressraum ein REXX-Programm läuft, kann es bestimmte Umgebun-
gen erreichen oder nicht, wobei es grundsätzlich immer eine Standardumgebung gibt. Soll ein Befehl 
aus einem REXX-Programm an ein erreichbares Umfeld zur Ausführung übergeben werden, wel-
ches nicht das Standardumfeld ist, muss der Befehl an das gewünschte Umfeld adressiert werden. 
Die folgende Tabelle zeigt die möglichen Adressräume, in denen REXX-Programme laufen können, 
die erste genannte Systemumgebung ist die standardgemäße, alle weiteren Umgebungen können 
erreicht werden. 

Adressraum Umgebungen 

MVS/Batch MVS, LINK, ATTACH 

TSO/E native TSO, MVS, LINK, ATTACH 

TSO/E ISPF TSO, MVS, ATTACH, ISPEXEC, ISREDIT 

NETVIEW NETVIEW/VTAM, MVS 

1.16. Befehlsübergabe an die Systemumgebung 

Wir wollen hier nicht diskutieren, unter welchen Umständen Instruktionen, die nicht vom REXX-In-
terpreter ausgeführt werden, sondern an das Umfeld von REXX zur Ausführung übergeben werden, 
mit oder ohne Eingrenzung zwischen Apostrophe(') oder Quotes(") funktionieren oder nicht. Wir ge-
hen hier immer davon aus, dass Befehle, die nicht vom Interpreter ausgeführt werden, REXX-seitig 
als Literale zu hinterlegen sind. Dabei ist zu beachten, dass das Literal unterbrochen und wieder 
fortgesetzt werden muss, wenn innerhalb des Befehles REXX-Variable eingesetzt werden sollen. 
Da in der Standardumgebung von REXX ein Apostroph(') häufig als Teil eines Kommandos genutzt 
wird (immer, wenn TSO weiß, dass es sich bei einem Operanden um einen Dateinamen handelt, 
und diese Datei vollqualifiziert angegeben wird, muss das Voranstellen der Benutzerkennung des 
Anwenders verhindert werden. Der Name der Datei ist in diesem Fall zwischen Apostrophe(') einzu-
grenzen), gehen wir im weiteren Verlauf davon aus, dass TSO-Kommandos als Literale, eingegrenzt 
zwischen Quotes("), zu schreiben sind. Ich bitte an dieser Stelle alle Freaks und Minimalisten unter-
tänigst um Verzeihung (man würde sicher das eine oder andere Byte bei der Kodierung einsparen 
können). Wird ein Befehl vom Interpreter an das Umfeld zur Ausführung übergeben, wird auch die 
Ablaufsteuerung für den weiteren Programmverlauf auf das Umfeld übertragen. Nach erfolgter Aus-
führung des Befehles meldet die entsprechende Umgebung (TSO) den Return Code des ausgeführ-
ten Befehles an REXX. Der Interpreter hinterlegt diesen Return Code in der Variablen RC und über-
nimmt wieder die weitere Steuerung des Programmablaufes. 

SAY "Datei vollqualifiziert (ohne Hochkomma)" 

PARSE EXTERNAL DATEI 

"DEL '"datei"'" 

IF rc <= 8 THEN SAY "alles klar" 

ELSE SAY "Dateizugriff nicht möglich" 

Läuft REXX in einem TSO-Adressraum, ist das Standard-Environment TSO. In diesem Fall können 
ohne Adressierung TSO-Kommandos ausgeführt werden. Auch die TSO/E-REXX-Kommandos (Be-
fehle, für deren Ausführung letztlich TSO bemüht wird, die aber nur aus REXX ausführbar sind) wie: 
NEWSTACK, QSTACK, DELSTACK, MAKEBUF, QBUF, QELEM, DROPBUF, EXECIO und SUB-
COM sind verfügbar. 

1.17. Parsing (Das Zerlegen von Zeichenketten) 

Durch Parsing ist ein REXX-Programm in der Lage, Wortketten auf eine sehr komfortable Art in 
mehrere Variable zu zerlegen. Grundsätzlich ist dies natürlich mit einer der vielfältigen Funktionen 
von REXX möglich, aber die Variante PARSE drängt sich in vielen Fällen einfach auf. Sehen wir uns 
einige der gängigen Verfahren an (eine detaillierte Syntaxbeschreibung lest ihr bitte im Teil-2 »Syn-
taxbeschreibung« nach): 

1.17.1. PARSE EXTERNAL 

Weist die Eingabe des Anwenders über die Terminaltastatur einer oder mehreren Variablen zu. 

/* REXX *****************************************/ 

SAY "Geben Sie Ihren Namen an" 

PARSE EXTERNAL name 

Die Bildschirmeingabe wird in der Variablen NAME hinterlegt. 

/* REXX *****************************************/ 

SAY "Name und Alter ? (durch BLANK getrennt)" 

PARSE EXTERNAL name alter 

SAY "Hallo," name"," 

SAY "Sie sind" alter "Jahre alt." 

Vorausgesetzt, ein Bediener macht, was man von ihm erwartet (diese Annahme war schon so man-
chen Programmes vorzeitiges Ende), wird das erste Wort der Tastatureingabe in der Variablen 
NAME, das zweite Wort in der Variablen ALTER hinterlegt. Beide Werte können beliebig weiterver-
arbeitet werden. 

1.17.2. PARSE PULL 

Parsing über den Operanden PULL durchzuführen, ist sicher die meistgenutzte aller Varianten. Je 
nach Situation verhält sich die Anweisung aber unterschiedlich. Seit ihr euch im Programmablauf 
nicht absolut über bestimmte Dinge im Klaren, führt die Anweisung unter Umständen nicht zu dem 
von Ihnen erwarteten Ergebnis. Beliebt ist diese Form wegen der erlaubten, verkürzten Schreib-
weise. Soll eine Wortkette einer Variablen zugewiesen und dabei zu Großbuchstaben konvertiert 
werden, hieße die Syntax hierfür 

PARSE UPPER PULL variable. 

Zum gleichen Ergebnis kommen wir mit der Angabe PULL variable. Nach dem viele EDV-Ge-

schädigte sich zu Minimalisten entwickelt haben, steht diese Kurzform in der Beliebtheitsskala der 
Programmierer natürlich ganz oben. Nun aber zu den Tücken des Befehles: 

Für REXX steht im TSO-Adressraum ein dynamisch verwalteter Speicherbereich zur Verfügung - 
der Stack. In diesen Bereich können Datensätze durch Instruktionen oder durch Lesen aus einer 
Datei eingearbeitet werden. Details über den Stack wollen wir hier zunächst nicht erörtern. Wie auch 
immer, PULL prüft immer zuerst den Stack. Liegt dort ein Datensatz, wird er entnommen (der Satz 
verlässt den Bereich physisch) und in der hinter PULL angegebenen Variablen abgelegt. Ist der 
Stackbereich leer, wendet sich PULL an die Tastatur. Das Programm wartet auf Bedieneraktion. 
Tippt der Anwender beliebige Daten, werden sie bei <Enter> der Variablen hinter PULL zugewiesen. 

Unter dieser Voraussetzung verhält sich der PULL-Befehl identisch zu PARSE UPPER EXTERNAL 

variable. Die Versuchung, durch die Angabe PULL variable zum gleichen Ergebnis zu kom-

men ist verständlicherweise groß. Dies macht aber nur Sinn, wenn sicher ist, dass beim Lesen von 
der Tastatur keine Daten im Stack liegen. 

1.17.3. PARSE NUMERIC 

Über PARSE NUMERIC variable können die Standards für die Verarbeitung numerischer Werte 

ermittelt werden. Der Reihe nach werden die aktuellen Werte für DIGITS FUZZ und FORM in der 
Variablen hinterlegt (könnte beispielsweise sein: 9 0 SCIENTIFIC). Näheres ist im Kapitel »Befehls-
syntax« unter NUMERIC nachzulesen. 

1.17.4. PARSE SOURCE 

Diese Variante liefert eine Reihe interessanter Informationen über das laufende Programm. Details 
siehe im Kapitel Befehlssyntax unter PARSE SOURCE. Ein kleiner Vorgeschmack aber durch nach-
folgendes Beispiel 

/* REXX *****************************************/ 

PARSE SOURCE . . progname . 

SAY "Hi, Fan - es grüßt Dich Dein Programm:" , 

progname 

Das dritte Wort der Wortkette enthält den Namen des laufenden Programmes in Großbuchstaben 
und wird in der Variablen PROGNAME abgelegt. Diese Information ist sehr hilfreich, wenn ein grö-
ßerer Dialog Meldungen auszugeben hat, und der Name des Moduls, das die Meldung erzeugt, 
ebenfalls ausgegeben werden soll (Programmnamen direkt in das Programm zu schreiben ist ge-
fährlich. Sollte das Programm einmal umbenannt werden, gibt es viel Aufregung). 

1.17.5. PARSE VALUE 

PARSE VALUE ist eine komfortable Art, einen String zu zerlegen. Soll beispielsweise die Systemzeit 
in drei Variablen (Stunde, Minute und Sekunde) zerlegt werden, kann die Trennung des Zeit-Strings 
über das Erkennen der Doppelpunkte(:) innerhalb der Zeit erfolgen. Die Ausgabe der Systemzeit 
erfolgt im Format hh:mm:ss. 

/* REXX *****************************************/ 

PARSE VALUE TIME() WITH STD ':' MIN ':' SEC 

Die Systemzeit wird über die Funktion TIME() ermittelt. Die Zeichenfolge wird von links nach rechts 
gelesen. Alle Zeichen bis vor das definierte Trennzeichen (Doppelpunkt) werden der Variablen STD 
zugewiesen. Das Trennzeichen wird ignoriert. Nach dem Trennzeichen werden alle bis zum nächs-
ten Trennzeichen vorkommenden Daten der Variablen MIN zugewiesen, das Trennzeichen wird wie-
der ignoriert. Der Rest des durch TIME() gelieferten Strings wird in der Variablen SEC hinterlegt. 

1.17.6. PARSE VAR 

Die Aufteilung einer Wortkette, welche in einer Variablen liegt, ist auf unterschiedliche Arten möglich. 
Auch hier bietet das Parsing wohl den höchsten Komfort. Nehmen wir an 

TEXT="Sein, oder nicht sein" 

PARSE VAR TEXT WORT1 WORT2 WORT3 WORT4 

Variable Inhalt 

WORT1 Sein, 

WORT2 oder 

WORT3 nicht 

WORT4 sein. 

 

Das Komma, das direkt an das erste Wort angrenzt, wird bei der Zerlegung der Wortkette als Be-
standteil des ersten Wortes übernommen. Prinzipiell wäre hier auch möglich, ein bestimmtes Zei-
chen (nehmen wir am besten das Komma) zum Trennzeichen zu erklären. Das Trennzeichen muss 
in einer Variablen abgelegt werden und die Variable bei der Auftrennung der Wortfolge zwischen 
runde Klammern gesetzt sein. Das Ganze nochmal in einem Beispiel  

TEXT="Sein, oder nicht sein" 

PARSE VAR TEXT WORT1 "," WORT2 WORT3 WORT4 

Variable Inhalt 

WORT1 Sein 

WORT2 oder 

WORT3 nicht 

WORT4 sein. 

 

Das Komma wird bei Separation der Wortkette nicht übernommen, da es als Trennzeichen definiert 
wurde. 

Durch die Angabe definitiver Bytepositionen kann die Trennung einer Zeichenkette über Parsing 
auch dann durchgeführt werden, wenn dies auf der Wortbasis nicht möglich ist. Auch hierzu eine 
kleine Demonstration:  

: 

TEXT="Dasistdochmalwasanderes" 

PARSE VAR TEXT 1 WORT1 4 WORT2 7 WORT3 11 , 

                 WORT4 14 WORT5 17 WORT6 

: 

Variable Inhalt 

WORT1 Das 

WORT2 ist 

WORT3 doch 

WORT4 mal 

WORT5 was 

WORT6 anderes 

1.17.7. PARSE VERSION 

Kann die Versionsnummer und das Datum des REXX-Interpreters ermitteln. 

1.17.8. PARSE ARG 

Diese Variante erlaubt neben PARSE PULL eine verkürzte Schreibweise. So ist ARG var gleich-

bedeutend mit PARSE UPPER ARG var und bewirkt die Übernahme von Daten, die beim Pro-

grammstart mitgeliefert wurden. Die Daten werden dabei zu UpperCase konvertiert. Die Datenüber-
gabe kann auf drei verschiedene Arten erfolgen: 

Expliziter Programmaufruf 

EX TEST(DEMO1) 'Der Pferd hat vier Bein' EXEC 

EX 'SP1.TEST.EXEC(DEMO1)' 'Der Pferd hat vier Bein' EXEC 

Implizierter Programmaufruf 

%DEMO1 Der Pferd hat vier Bein 

Programmaufruf aus einem REXX-Programm 

CALL DEMO1 'Der Pferd hat vier Bein' 

Alle diese Varianten bedingen, dass im Programm DEMO1 die Wortkette "Der Pferd hat vier Bein" 
aufgenommen wird. Wie dies auszusehen hat, sehen wir uns am besten in einigen Beispielen an: 

/* REXX *****************************************/ 

ARG DATEN 

Der Inhalt der Variablen DATEN wäre bei der verkürzten Schreibweise (gleichzusetzen mit PARSE 
UPPER ARG DATEN) die Wortkette zu Großschreibung konvertiert. 

Variable Inhalt 

DATEN DER PFERD HAT VIER BEIN 
 

/* REXX *****************************************/ 

ARG WORT1 WORT2 WORT3 WORT4 WORT5 

Hier wird die Wortkette auf Wortbasis zerlegt. Das erste Wort wird der ersten Variablen hinter ARG 
zugewiesen, das zweite Wort der zweiten Variablen, und so weiter. 

Variable Inhalt 

WORT1 DER 

WORT2 PFERD 

WORT3 HAT 

WORT4 VIER 

WORT5 BEIN 
 

/* REXX *****************************************/ 

ARG WORT1 WORT2 WORT3 

Gehen wir vom gleichen Startaufruf aus. Zunächst erfolgt wieder eine Zuweisung Wort nach Variab-
len von links nach rechts. Sind weniger Variablen hinter ARG angegeben als Worte im Startaufruf 
übergeben wurden, führt dies nicht zum Datenverlust, sondern alle überzähligen Worte werden in 
der letzten Variablen gesammelt. 

Variable Inhalt 

WORT1 DER 

WORT2 PFERD 

WORT3 HAT VIER BEIN 
/* REXX *****************************************/ 

ARG WORT1 WORT2 WORT3 WORT4 WORT5 WORT6 WORT7 

: 

Ebenso könnten in der ARG-Instruktion mehr Variablen vorgesehen sein, als übergeben werden. 
Für die überzähligen Variablen erfolgt eine NULLSTRING-Zuweisung. 

Variable Inhalt 

WORT1 DER 

WORT2 PFERD 

WORT3 HAT 

WORT4 VIER 

WORT5 BEIN 

WORT6 Nullstring 

WORT7 Nullstring 

/* REXX *****************************************/ 

PARSE ARG WORT1 WORT2 WORT3 WORT4 WORT5 

Wird die Klausel PARSE kodiert, wird der Befehl nicht mehr in seiner ursprünglichen Form (PARSE 
UPPER ARG) interpretiert. Die Konvertierung der Variablen auf Großschreibung unterbleibt. 

Variable Inhalt 

WORT1 Der 

WORT2 Pferd 

WORT3 hat 

WORT4 vier 

WORT5 Bein 

1.17.9. Der Punkt als Begrenzungszeichen 

Häufig ergibt sich das Problem, dass ein Programm nur eine begrenzte Anzahl von Parametern 
übernehmen soll, der Anwender aber bei Programmaufruf seine ganze Lebensgeschichte liefert. 
Alle überzähligen Worte würden in der letzten Variablen hinter ARG gesammelt werden. Gehen wir 
davon aus, dass zwei Worte übernommen werden sollen, der unter Umständen gelieferte Rest soll 
nicht berücksichtigt werden. Dieses Problem kann gelöst werden, indem ein Begrenzungszeichen 
die Zuweisung des Parsings beendet. 

/* REXX *****************************************/ 

PARSE ARG WORT1 WORT2 . 

Variable Inhalt 

WORT1 Der 

WORT2 Pferd 

1.17.10. Der Punkt als Platzhalter 

Der Punkt kann auch als Platzhalter bei der Wertzuweisung des Parsings eingesetzt werden. Erfolgt 
die Zuweisung der einzelnen Parameter-Worte zu den der Reihe nach angegebenen Variablen, kann 
der Punkt an die Stelle einer Variablen treten. Das dazu analoge Parameterwort wird bei der Zuwei-
sung entsprechend ignoriert. 

/* REXX *****************************************/ 

ARG WORT1 . WORT2 . WORT3 

Variable Inhalt 

WORT1 Der 

WORT2 hat 

WORT3 Bein 

 

oder 

/* REXX *****************************************/ 

ARG WORT1 . . WORT2 . 

Variable Inhalt 

WORT1 Der 

WORT2 vier 

Nachdem wir nun mehrfach einen kleinen Einzeiler als Demonstration für Datenübergabe an ein 
Programm benutzt haben, hier die vollständige Fassung des Gedichtes von Heinz Erhardt: 

Der Pferd hat vier Bein,  
an jedem Eck ein.  
Wenn Pferd drei Bein hätt’ -umfall’n tät’. 

1.17.11. Parse für Fortgeschrittene 

Parsing kann (auch) auf Grund von Trennzeichen oder Bytepositionen erfolgen. Dabei ist zu beach-
ten: 

▪ Soll das Trennzeichen über eine Variable geliefert werden, so ist diese in Klammern zu ko-
dieren 

▪ Soll die Byteposition über eine Variable geliefert werden, ist diese in Klammern zu setzen 
und mit einem führenden Gleichheitszeichen zu kodieren. 

/* REXX 

*/ 

A=date("E") 

B="/" 

C=lastpos("/",A) 

Parse var a Tag (B) Monat =(C) Jahr 

exit 

Das folgende Programbeispiel soll in verschiedenen Subroutinen zeigen, wie mächtig der Parse-
Befehl ist. Meiner Meinung nach ist ein REXX-Programm ohne Parse-Instruktion kein REXX-Pro-
gramm 

/* REXX 

 

   Diverse Beispiele für PARSE 

 

   Wer PARSE nicht benutzt, hat REXX nicht kapiert, oder: 

   Nicht alles was REXX-Syntax hat, ist ein REXX-Programm 

 

   Bei der Ausführung zuschauen? Einfach den TRACE einschalten 

trace ?r 

*/ 

call sample1 

call sample2 date() 

call sample2 date("E") 

call sample2 "21.02.2014" 

call sample3 

call sample4 

call sample5 

call sample6 

call sample7 

call sample8 

exit 

sample1: procedure 

   /*          ----+----1----+----2----+----3----+----4----+----5----+*/ 

   Satz     = "Paul Hugo Meier     12345 Irgendwo      Dorfplatz 5" 

   StrtPos  = "1         11        21    27            41" 

   parse var StrtPos BegName BegVName BegPlz BegOrt BegStr . 

   parse var satz =(BegName)  Name    , 

                  =(BegVName) Vorname , 

                  =(BegPlz)   Plz     , 

                  =(BegOrt)   Ort     , 

                  =(BegStr)   Strasse 

Return 

 

sample2: procedure 

   arg datum 

   select 

      when words(datum)  = 3 then dlm = " " 

      when length(datum) = 8 then dlm = "/" 

      otherwise                   dlm = "." 

   end 

   parse var datum Tag (dlm) Monat (dlm) jahr 

return tag"-"monat"-"jahr 

 

sample3: procedure 

   parse value "" with a b c d e 

return 

 

sample4: procedure 

   parse value "A-Wert~Das ist B~~123~" with , 

                A "~" B "~" C "~" D "~" E 

Return 

 

sample5: procedure 

   dmy = "----+----1----+----2----+----3----+----4----+----5---" 

   v.1 = "Seppl Moser  Jack Ripper    Krimi    Auge der Nacht" 

   v.2 = "Jo    Napf   Nachtschatten  Thriller Windjammer" 

   v.3 = "Hein  Doof   Blödian        Fachbuch Intelligenz-Allergie" 

   do §=1 to 3 

      parse var v.§ VName  7 NName 14 Pseudonym  29 Genre 38 Titel 

      parse var v.§ VName +6 NName +7 Pseudonym +15 Genre +9 Titel 

   end 

return 

 

sample6: procedure 

   dmy  = "----+----1----+----2----+----3----+----4----+----5---" 

   data = "Seppl Moser   12345 St. Nimmerlein  Wart ein Weil 3" 

   parse var data VN  7 NN 15 PLZ 21 Ort 37 Str 

   parse var data VN +6 NN +8 PLZ +6 Ort +16 Str 1 VN_NN +14 Adr 

return 

 

sample7: procedure 

   bsp = "1234567890" 

   parse var BSP First5 +5 SixAnd7 +2 -5 ThreeAnd4 +2 + 4 Last2 

return 

sample8: procedure 

   delim = "41"x 

   parse value "5 7 Gaga"delim"Das steht in der Variablen Hugo" !! , 

      delim"August"delim !! date() !! delim , 

      with a b gaga (delim) hugo (delim) august (delim) datum (delim) . 

 

   say "A       " a 

   say "B       " b 

   say "GAGA    " gaga 

   say "HUGO    " hugo 

   say "AUGUST  " august 

   say "DATUM   " datum 

return 

1.17.12. Parse für die Ausgebufften 

Es kommt immer wieder die Frage auf, ob man REXXe nicht mit Default Parametern versehen kann, 
die gelten, wenn die entsprechenden Parameter beim Aufruf nicht übergeben werden. Generell erst-
mal NEIN. Andererseits heißt es ja immer: Geht nicht gibt's nicht. Wenn die REXX als Function 
aufgerufen wird, lässt sich das realisieren (an dieser Stelle besten Dank an Christian Zitzmann für 
dieses Beispiel, Dank auch an Prof. Rony Flatscher von der Wirtschaftsuni in Wien, der die Anregung 
dazu gegeben hat): 

/* REXX * Parsing mit Defaults. Tut nur in Functions */ 

parse value parse_fkt(,"ADRTEST") with datei member ziel 

 

say "Datei " datei      /* SP1.EDU.DATA */ 

say "Member" member     /* ADRTEST      */ 

say "Ziel  " ziel       /* MVS1         */ 

exit 

 

parse_fkt: procedure 

   parse value arg(1) "SP1.EDU.DATA" "~" , 

               arg(2) "ADR1"         "~" , 

               arg(3) "MVS1"         "~" , 

               with DSN . "~" MBR . "~" Target . "~" 

return dsn mbr target 

1.18. Testhilfen, Debugging, Eventhandling 

Es gibt nicht umsonst eine Reihe mehr oder minder intelligente Aussprüche wie: Ein Programm 
macht was man kodiert, nie was man will! Leistungsfähige, moderne Hochsprachen wie REXX 
verleiten dazu, im Hau-Ruck-Stil zu programmieren (erst mal alles hinschreiben, was einem so ein-
fällt; Programm laufen lassen; angezeigte Fehler korrigieren; neuer Versuch). Vieles lässt sich auf 
diese Weise bewältigen. Manche Fehler sind aber etwas komplizierter. Tritt ein Fehler in einem 
Statement als Folge einer falschen Zuweisung in einem vorangegangenen Befehl auf, oder handelt 
es sich nicht um syntaktische Mängel im Programm, sondern um logische Fehler, die zu einem an-
deren als dem gewünschten Programmverlauf führen, ist die Suche nach der Ursache oft nicht so 
einfach. REXX bietet an dieser Stelle einen Ablaufverfolger in Form des Befehles TRACE (eine de-
taillierte Beschreibung der Möglichkeiten und Syntax des Befehles findet ihr im Teil-2 »Syntaxbe-
schreibung«). Die gängigste Form, den Ablaufverfolger zu starten, ist die Angabe TRACE "?R". 

Man sieht in der Praxis häufig eine leicht abartige Vorgehensweise im Umgang mit TRACE. Läuft 
ein Programm nicht wie es soll, wird der TRACE-Befehl in das Programm kodiert. Testlauf. Man 
glaubt, den Fehler gefunden zu haben, korrigiert ihn und löscht den TRACE-Befehl wieder. Nächster 
Versuch - neuer Fehler. Einfügen des TRACE-Befehles. Programmstart. Fehlererkennung - Korrek-
tur - TRACE löschen... 

Sinnvoller ist, das Programm so zu schreiben, dass das Aktivieren des Ablaufverfolgers über die Art 
des Programmaufrufes geschehen kann. Wir gehen im nachfolgenden Beispiel davon aus, dass das 
Programm sich normal verhält, wenn bei Programmstart keine Parameter angegeben werden. Wird 
bei Programmaufruf als Parameter dagegen ein T (wie Test) übergeben, soll der Ablaufverfolger 
eingeschaltet werden. 

/* REXX  

*/ 

ARG DEBUG . 

IF ABBREV("TEST",DEBUG,1) THEN TRACE "?R" 

Im Modus ?R ist dem Anwender während des Programmablaufs die Kommunikation mit dem Inter-
preter möglich. Die Befehle werden interaktiv ausgeführt. Der Interpreter zeigt einen Befehl, führt ihn 
aus, teilt das Ergebnis des Befehles mit und setzt einen Haltepunkt. Am Haltepunkt bestimmt der 
Anwender, wie es weitergeht. Drückt der Anwender <ENTER>, ohne Daten einzugeben, geht der 
Interpreter an die Ausführung des nächsten Programmbefehles in gleicher Weise. Nach Befehlsaus-
führung setzt der Interpreter wieder einen Haltepunkt und der Anwender ist wieder am Zug. Prinzi-
piell ist dem Anwender am Haltepunkt jede beliebige Befehlseingabe möglich, die der Interpreter 
verarbeiten kann. Dies können REXX-Instruktionen oder Befehle an ein erreichbares Umfeld von 
REXX sein. Solange der Anwender mit <Enter> eine Eingabe übergibt, versucht der Interpreter diese 
Eingabe als Befehl auszuführen, ohne mit der eigentlichen Programmverarbeitung fortzufahren und 
setzt aufs Neue einen Haltepunkt. Der Anwender hat so die Möglichkeit, während des Programmab-
laufes die Inhalte von Variablen abzufragen und gegebenenfalls temporär im Ablauf zu verändern. 
Generell ist ihm jeder Befehl möglich. Wird ein Fehler erkannt, der eine weitere Programmverarbei-
tung sinnlos macht, wäre eine mögliche Aktion des Anwenders auch die Eingabe des Befehles EXIT, 
der zum unmittelbaren Programmende führen würde. Achtung bei Dialoganwendungen: Alle Tabel-
len bleiben geöffnet, alle File Tailoring Outputs verbleiben im virtuellen Speicher. 

Wird im Trace beispielsweise erkannt, dass eine Variable falsch besetzt ist und aufgrund dessen ein 
Befehl nicht ordnungsgemäß ausgeführt werden kann, wäre möglich, am Haltepunkt die Variable 
korrekt zu definieren und nach <ENTER> nur ein Gleichheitszeichen(=) einzugeben. Dies fürht zur 
Wiederholung des letzten im Programm ausgeführten Befehls. 

An dieser Stelle müssen wir uns kurz die standardgemäße Fehlerbehandlung des Interpreters an-
sehen. Tritt beispielsweise ein Syntaxfehler auf, meldet der Interpreter sehr präzise, was ihm nicht 
passt (ich frage mich oft, warum er den Fehler nicht gleich selbst korrigiert, wenn er ihn schon so 
differenziert erkennen kann). Bei Syntaxfehler liefert der Interpreter die Nummer der Zeile aus dem 
Quellencode, die zum Fehler führte, zeigt deren Inhalt und gibt eine Fehlernummer (RC) und die 
dazugehörende Fehlermeldung aus. Was will man mehr erwarten? Nun, demonstrieren wir es in 
einem Beispiel 

/* REXX */ 

ARG TEST . 

IF TEST = "T" THEN TRACE "?R" 

A="BlaBla" 

B=3 

C=7 

D=5 

ERGEBNIS = A + B * C - D 

SAY "Ergebnis:" ERGEBNIS 

EXIT 

Der Interpreter zeigt beim Ablauf des Programmes folgende Meldung:  

8 +++ ERGEBNIS = A + B * C - D 

IRX0041I Error running PROG1, line 8: 

Bad arithmetic conversion 

und beendet das Programm unmittelbar. Die gemeldete Ursache (Bad arithmetic conversion) liegt 
in einem Fehler bei der Umsetzung der Variablen in der Befehlszeile. Mindestens eine davon hat 
keinen numerischen Inhalt. Die einzige Variable, die nicht Verursacher des Problems sein kann, ist 
ERGEBNIS, da sie mit dem Resultat der Auflösung des Ausdruckes rechts vom Gleichheitszeichen 
hätte gefüllt werden sollen. Welche der Variablen A, B, C oder D aber Grund für den Abbruch war 
(es können auch mehrere oder alle Variablen falsch gefüllt sein), bleibt verborgen. Werden die Va-
riablen an unterschiedlichsten Stellen des Programmes gebildet und ist ihr jeweiliger Inhalt situati-
onsabhängig, ist die Suche nach der Ursache oft mühsam. Es wäre für uns eine große Hilfe, wenn 
nach Auftreten des Fehlers, vor Programmende, die Inhalte der Variablen abgefragt werden könn-
ten. Fehlerbehebung wäre eine Kleinigkeit. Hier ist nun angesagt, die Fehlerbehandlung selbst in 
die Hand zu nehmen. Dies hat den bitteren Beigeschmack, dass sich der Interpreter um nichts mehr 
kümmert (Quellencodezeilennummer und -inhalt sowie eine Fehlerursache werden nicht gemeldet). 
Was soll’s, machen wir es eben selbst (Es gibt viel zu tun, gehen wir heim).  

SIGNAL heißt das Zauberwort. Über die SIGNAL-Anweisung können wir den Interpreter veranlas-
sen, bei Auftreten einer bestimmten Situation (in unserem Fall SYNTAXFEHLER) ein Sprungziel 
innerhalb des Programmes anzusteuern. Was dort zu geschehen hat, können wir selbst veranlas-
sen. Sinnvollerweise muss das Sprungziel der Fehlerbehandlung hinter dem regulären Programm-
ende liegen, damit es bei normalem Ablauf nicht angesteuert wird. 

/* REXX */ 

ARG TEST .  

IF TEST = "T" THEN TRACE "?R" 

SIGNAL ON SYNTAX NAME FEHLER 

A="BlaBla" 

B=3 

C=7 

D=5 

ERGEBNIS = A + B * C - D 

SAY "Ergebnis:" ERGEBNIS 

EXIT 

 

FEHLER: 

PARSE SOURCE . . progname . 

SAY ">>" progname "Zeile" SIGL":" SOURCELINE(SIGL) 

SAY ">> RC:" RC "-" ERRORTEXT(RC) 

TRACE "?R" 

NOP 

EXIT 

Tritt nun im Ablauf der Syntaxfehler auf, kümmert sich der Interpreter nicht um Ursachenforschung, 
sondern verzweigt zum Label FEHLER: und führt die Befehle aus, die dort kodiert stehen. Der 
PARSE-Befehl ermittelt den Namen des laufenden Programmes und hinterlegt ihn in der Variablen 
PROGNAME. Die speziellen Variablen RC und SIGL enthalten den Return Code und die Zeilen-
nummer des fehlerverursachenden Befehles (SIGL ist die Zeilennummer eines Befehles im Quel-
lencode, der aufgrund einer Sprungbedingung (SIGNAL) verlassen wurde. Wurde der Sprung durch 



Inside z/OS Seite 38 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

SIGNAL ON SYNTAX herbeigeführt, ist es der Befehl, in dem der Syntaxfehler auftrat). Der Syn-
taxfehler meldet sich für den Anwender mit folgender Optik: 

>> PROG1 Zeile 9: ERGEBNIS = A + B * C - D 

>> RC: 41 - Bad arithmetic Conversion 

Hier ist das Programm aber nicht unmittelbar zu Ende, die Ablaufsteuerung behält der Interpreter. 
Der Ablaufverfolger wird eingeschaltet und in den interaktiven Modus versetzt. Bei dieser Gelegen-
heit eine Charakteristik des Trace-Kommandos und ein neuer Befehl: Nachdem der Interpreter den 
Quellencode Zeile für Zeile liest und interpretiert, wird der Ablaufverfolger nicht in der Zeile TRACE 
aktiv (der Interpreter weiß erst was zu tun ist, wenn er die Zeile gelesen hat), sondern zusammen 
mit dem nächsten Befehl. Wäre dieser nächste Befehl nun EXIT (Programmende), wäre dies aus-
gesprochenes Pech. Der Ablaufverfolger könnte nicht aktiviert werden. Letztlich spielt es keine Rolle, 
wie der nächste Befehl heißt, um den Ablaufverfolger einzuschalten, außer EXIT oder RETURN. An 
dieser Stelle bietet sich eine Instruktion, die ihrerseits nichts tut, direkt an (NOP für No OPeration). 
Diese Instruktion genügt, um den Ablaufverfolger zu aktivieren und einen Programmhaltepunkt zu 
setzen. Jetzt ist der Anwender an der Reihe. Solange am Haltepunkt nicht nur <ENTER> gedrückt 
wird, versucht der Interpreter die Eingabe als Befehl auszuführen und bleibt in der Programmverar-
beitung stehen. Um die problemverursachende Variable zu erkennen ist jetzt nur die Eingabe SAY 

A B C D nötig. Die Inhalte der Variablen werden am Bildschirm ausgegeben. Man kann unmittelbar 

feststellen, wo die Ursache des Problems liegt. Wenn keine Abfragen mehr durchgeführt werden 
sollen, drückt der Anwender <ENTER>, ohne Daten einzugeben. Der Interpreter führt den nächsten 
Befehl aus (EXIT) und beendet das Programm. Gleichgültig, ob es theoretisch möglich wäre, wieder 
in das Programm an die fehlerverursachende Stelle zurückzukehren oder nicht, es gehört sich ein-
fach nicht. 

Wird im Ablauf eines REXX-Programmes <PA-1> gedrückt (ATTENTION-INTERRUPT-KEY), unter-
bricht der Interpreter zunächst die Verarbeitung und erlaubt dem Anwender eine Eingabe. An dieser 
Stelle wäre möglich: 

Eingabe Bedeutung 

HI Halt Interpretation 
Programmende, siehe auch SIGNAL ON HALT 

HT Halt Typing  
Bildschirmausgaben werden unterdrückt 

RT Resume Typing 
Bildschirmausgaben werden wieder angezeigt 

TE Trace End 
Der Trace-Modus wird beendet 

TS Trace Start 
Startet den Ablaufverfolger 

1.18.1. Empfehlung 

Die einfachste und effektivste Art, ein komplexes Programm auf Fehler hin zu prüfen, ist der Com-
pile! Natürlich werden auch hier nicht alle Fehler gefunden aber die meisten. 

1.18.2. Eventbehandlung 

Das folgende Beispiel zeigt eine standardisierte Eventbehandlungsroutine, die in jedes Programm 
eingebunden werden kann. Welche der Events letztlich die Definition per SIGNAL ON oder CALL 
ON aktiviert werden sollen, entscheidet der Programmierer selbst. Bedenkt aber, dass ein SIGNAL 
ON ERROR auf jeden Return Code > 0 reagiert und damit vielfältige Probleme aufwerfen kann. Es 

gibt einige Befehle, welche über RC eine Information zurückgeben und wenn dieser nicht 0 ist, wäre 
das grundsätzlich kein Fehler (z.B.: QSTACK, QELEM). 

/* REXX */ 

parse source . RxTyp RxName . 

/*  

Signal on Error Name Event_Handler 

Signal on Failure Name Event_Handler 

*/ 

Signal on NoValue Name Event_Handler 

Signal on Syntax  Name Event_Handler 

 

call msg "OFF" 

 

Main_Start: 

/* Was auch immer kodiert werden soll 

*/ 

Main_Ende: 

Exit 

 

Event_Handler: 

   /****************************************************************** 

   * Behandlung diverser Events                                      * 

   ******************************************************************/ 

   parse source . . RxN . 

   CondC = Condition("C") 

   CondD = Condition("D") 

   CondS = Condition("S") 

   CondI = Condition("I") 

   Select 

      when CondC = "NOVALUE" then do 

         say RxName"-E" "NOVALUE Fehler in Zeile" , 

             SigL", undefinierte Variable: "CondD 

      End 

      when CondC = "SYNTAX" then do 

         say RxName"-E" "SYNTAX Fehler in Zeile "SigL 

         say RxName"-E" "RC = "rc", "errortext(rc) 

         say RxName"-E" "Desc: "CondD 

      end 

      when wordpos(CondC,"ERROR FAILURE") > 0 then do 

         say RxName"-E" CondC": RC = "rc" in Zeile "SigL 

         say RxName"-E" "Statement: "CondD 

      end 

      otherwise do 

         say RxName"-E" CondC"-Fehler in Zeile "SigL 

         say RxName"-E" "COND(D): "CondD 

         say RxName"-E" "COND(S): "CondS 

         say RxameN"-E" "COND(I): "CondI 

         say RxameN"-E" "Return Code = "rc 

      end 

   end 

   if sourceline() > 0 then, 

      say right(SigL,6)": "left(sourceline(SigL),72) 

   say RxN"-E" "Rexx endet mit RC = 16" 

   if SYSVAR("SYSISPF") = "ACTIVE" then do 

      zispfrc = 16 

      address ispexec "VPUT (ZISPFRC) SHARED" 

   end 

exit 16 

1.19. Dialog mit dem Anwender 

Soll ein REXX-Programm während seines Ablaufes mit dem Anwender kommunizieren, stehen hier-
für zwei Befehle zur Verfügung: SAY und PARSE. 

Gleich vorweg: Formatierte Bildschirmein-/ausgaben über Zeilen/Spalten-Koordinaten sind in REXX 
nicht möglich (geht nicht, gibt's nicht – ist aber mit regulären REXX-Mitteln nicht machbar). 

SAY gibt eine Zeile am Terminal aus. Jeder SAY-Befehl erzeugt eine neue Zeile. Sollen nur Leer-
zeilen erzeugt werden, wird hinter SAY nichts kodiert. Zeichenketten und Variable können in loser 
Folge im SAY-Befehl ausgegeben werden. 

PARSE erlaubt das Lesen von der Tastatur und weist die Eingabe einer oder mehreren Variablen 
zu. 

/* REXX  

*/ 

SAY "Geben Sie Ihren Namen ein" 

PARSE EXTERNAL name 

SAY "Geben Sie Ihr Alter an" 

PARSE EXTERNAL alter 

SAY "Geben Sie Ihre Körpergröße in cm an" 

PARSE EXTERNAL groesse 

SAY "Stellen Sie sich aufrecht vor das Terminal" 

SAY "Durch mein eingebautes Infrarot-Auge" 

SAY "muss ich Ihre Größe prüfen" 

SAY "Nachdem Sie sich wieder hingesetzt haben," 

SAY "drücken Sie bitte auf ENTER" 

PARSE EXTERNAL 

SAY "Ergebnis nach eingehender Prüfung:" 

SAY "Sie sind höchstens" groesse-3 "cm groß." 

SAY "3 Kilo weniger würden Ihnen sicher guttun..." 

: 

1.20. Abfragen und Abfragetechniken 

Eines der wichtigsten Sprachelemente in jeder Programmiersprache ist die Entscheidungsfindung. 
REXX bietet zwei verschiedene Möglichkeiten, eine beliebige Situation zu hinterfragen. Prinzipiell 
wäre eine Abfrageform zwar ausreichend, um jedes noch so komplizierte Problem zu lösen. Wir 
werden aber anhand einiger Beispiele erkennen, dass die zwei Varianten in REXX sehr viel Komfort 
in die Kodierung bringen und jede von ihnen ihre Berechtigung hat. 

1.20.1. Einfache Abfragen (IF-THEN-ELSE) 

Die einfache Abfrage ist immer dann angebracht, wenn eine Frage mit einem klaren JA(THEN) oder 
NEIN(ELSE) zu beantworten ist. Wird die Bedingung der Abfrage bejaht, wird der Befehl hinter 
THEN ausgeführt. Die Kodierung des ELSE-Zweiges ist syntaktisch nicht erforderlich, wenn sie von 
der Ablauflogik her nicht benötigt wird. 

SAY "Gib eine Zahl an" 

PULL zahl 

IF zahl // 2 = 0 THEN SAY zahl "ist gerade" 

/* Weitere Verarbeitung */ 

 

SAY "Gib eine Zahl an" 

PULL zahl 

IF zahl // 2 = 0 THEN SAY zahl "ist gerade" 

ELSE SAY zahl "ist ungerade" 

/* Weitere Verarbeitung */ 

Grundsätzlich gilt für beide Zweige: Es darf nur ein Befehl an die JA- oder NEIN-Bedingung geknüpft 
werden. Sollen abhängig von einer Bedingung mehrere Befehle ausgeführt werden, sind sie zu einer 
Befehlsfolge zusammenzufassen, die zwischen DO und END eingegrenzt wird. 

SAY "Gib Dein Alter an" 

PULL alter 

IF DATATYPE(alter) = "CHAR" THEN DO 

   SAY "Eingabe nicht numerisch" 

   SAY "Vorzeitiges Programmende" 

   EXIT 

END 

ELSE DO 

   SAY "Bravo" 

   /* Weitere Verarbeitung */ 

END 

Komplexere Probleme lassen sich durch simple Abfragen oft nicht lösen. Grundsätzlich kann ein IF-
Konstrukt eine sehr komplexe Abfragelogik enthalten (Boolesche Verknüpfungen), die Lesbarkeit 
wird dadurch aber nicht gefördert. Wir wollen versuchen eine Datumsprüfung als abschreckendes 
Beispiel über ein IF-Gebilde durchzuführen und bedienen uns hierzu dreier Varianten. 

Das erste Beispiel zeigt eine Folge von einfachen Abfragen. Nachteil hierbei ist: Wenn eine Fehler-
bedingung erkannt ist, hat eine weitere Prüfung im Grunde keinen Sinn mehr und belastet nur unnütz 
die Systemleistung: 

/* REXX */ 

SAY "Gib Dein Geburtsdatum an (Format: tt.mm.jj)" 

PULL gebdat 

PARSE VAR gebdat 1 tt 3 x1 4 mm 6 x2 7 jj 

IF LENGTH(gebdat) >< 8 then say "Falsche Länge" 

IF x1!!x2 >< ".." THEN SAY "Falsches Format" 

IF DATATYPE(tt!!mm!!jj) = "CHAR" THEN 

   SAY "Falsches Format" 

IF mm < 01 ! mm > 12 THEN SAY "Monat falsch" 

IF POS(mm,"01 03 05 07 08 10 12") > 0 THEN tt_max=31 

IF POS(mm,"04 06 09 11") > 0 THEN tt_max = 30 

IF mm = 02 THEN IF jj // 4 = 0 THEN tt_max = 29 

ELSE tt_max = 28 

IF tt > tt_max THEN SAY "Tag falsch" 

Angenommen, der Anwender hätte 17.aaa.1970 eingetippt, würden wegen der voneinander unab-
hängigen Prüfungen folgende Meldungen ausgegeben werden: 

Falsche Länge      /* Länge nicht acht Byte */ 

Falsches Format    /* Byte 3 und 6 nicht .. */ 

Falsches Format    /* Tag/Monat/Jahr nicht numerisch */ 

Monat falsch       /* mm ist kleiner 01 */ 

Tag falsch         /* TT_MAX wegen falschen Monats nicht 

                      Gefüllt, zeigt auf sich selbst, ist 

                      somit CHAR, TT>TT_MAX ist true */ 

Versuchen wir es mal mit einem in sich geschachtelten IF-THEN-ELSE-Konstrukt: 

SAY "Gib Dein Geburtsdatum an (Format: tt.mm.jjjj)" 

PULL gebdat 

PARSE VAR gebdat 1 tt 3 x1 4 mm 6 x2 7 jjjj 

IF LENGTH(gebdat) >< 8 then say "Falsche Länge" 

ELSE 

   IF x1!!x2 >< ".." THEN SAY "Falsches Format"  

   ELSE 

      IF DATATYPE(tt!!mm!!jjjj) = "CHAR" THEN , 

         SAY "Falsches Format" 

      ELSE 

         IF mm < 01 ! mm > 12 THEN SAY "Monat falsch" 

         ELSE DO 

         IF POS(mm,"01 03 05 07 08 10 12") > 0 THEN , 

            tt_max=31 

         ELSE 

            IF POS(mm,"04 06 09 11") > 0 THEN , 

               tt_max = 30 

            ELSE 

               IF mm = 02 THEN , 

                  IF jjjj // 4 = 0 THEN tt_max = 29  

                  ELSE tt_max = 28 

               IF tt > tt_max THEN SAY "Tag falsch" 

END 

Unter gleicher Voraussetzung wird hier nur noch falsches Format aufgrund der Längenprüfung ge-
meldet. Alle anderen Prüfungen werden übersprungen, da die jeweils nachfolgende Prüfung nur 
erfolgt, wenn die vorangegangene Prüfung ein logisches FALSE(unwahr) geliefert hat. Aber ent-
scheidet selbst. Schön und lesbar ist etwas anderes. 

Für die Enthusiasten der Booleschen Verknüpfungen wäre auch die folgende Lösung denkbar: 

SAY "Gib Dein Geburtsdatum an (Format: tt.mm.jj)" 

PULL gebdat 

PARSE VAR gebdat 1 tt 3 x1 4 mm 6 x2 7 jj 

IF LENGTH(gebdat) >< 8 ! x1!!x2 >< ".." ! , 

   DATATYPE(tt!!mm!!jj) = "CHAR" THEN 

      SAY "Falsches Format" 

ELSE 

   IF mm < 01 ! mm > 12 THEN SAY "Monat falsch" 

   ELSE 

      IF POS(mm,"01 03 05 07 08 10 12")>0 & tt > 31 ! , 

         POS(mm,"04 06 09 11") > 0 & tt > 30 ! , 

         mm = 02 & jj // 4 = 0 & tt > 29 ! , 

         mm = 02 & jj // 4 > 0 & tt > 28 THEN , 

            SAY "Tag falsch" 

Die Änderungsfreundlichkeit bei dieser Version ist gleich Null. Durch die mehrfache Verknüpfung 
von Ausdrücken bleibt die Lesbarkeit eines Programmes häufig auf der Strecke. Im Ablauf des Pro-
grammes liefern die Verknüpfungen keinerlei Vorteile, da eine komplexe, verknüpfte Abfrage intern 
vom Interpreter in mehrere einzelne Abfragen zerlegt wird. Darüber hinaus ist bei Verknüpfungen 
eine gezielte Meldungsausgabe nicht möglich (warum ist die Tagesangabe falsch). Beurteilungskri-
terium, ob ein Programm gut oder schlecht ist, ist ganz sicher nicht, ob es funktioniert oder nicht. 
Dass ein Programm ein richtiges Ergebnis liefert, ist eine Grundbedingung und das Mindeste, was 
man erwarten kann. Erfüllt es noch nicht mal diese Bedingung, ist es kein Programm, sondern 
höchstens ein Versuch. Im Vordergrund stehen die Lesbarkeit, die Änderungsfreundlichkeit, die Ab-
laufgeschwindigkeit und die Größe des Programmes, in genannter Reihenfolge. 

1.20.2. Mehrfachabfragen (SELECT) 

Mehrfachabfragen sind sinnvoll, wenn zur Entscheidungsfindung eine Fragestellung mit einem simp-
len JA/NEIN nicht ausreicht. Im Grunde funktioniert eine Mehrfachabfrage wie ein verschachteltes 
IF-THEN-ELSE-Konstrukt, ist aber wesentlich einfacher zu lesen. Wie bei der IF-THEN-ELSE-Ver-
schachtelung wird die zweite Frage im SELECT nur gestellt, wenn die erste Frage verneint wurde. 
Die dritte Frage wird nur berücksichtigt, wenn die ersten beiden Fragen negativ beantwortet wurden, 
und so weiter. Trifft eine Fragestellung zu, erfolgt die vorgesehene Aktion. Auch hier gilt: An das 
Bejahen einer Frage kann nur ein Befehl geknüpft werden. Sollen mehrere Befehle abhängig vom 
JA einer Frage ausgeführt werden, sind sie als Befehlsfolge zwischen DO und END einzugrenzen. 

Ist die Aktion des JA-Falles durchgeführt, wird das Mehrfachabfragengebilde verlassen, ohne dass  
weitere Fragestellungen berücksichtigt werden. 

Auch für die Mehrfachabfrage sollte ein NEIN-Zweig kodiert werden. Dieser NEIN-Zweig wird durch-
laufen, wenn von allen gestellten Fragen innerhalb der Mehrfachabfrage keine zutreffend war. Tritt 
dieser Fall ein und ist kein NEIN-Zweig kodiert, führt dies zum Programmabbruch mit Syntaxfehler. 

In REXX wird die Mehrfachabfrage mit dem Befehl SELECT eingeleitet und durch END abge-

schlossen. Die einzelnen Fragen werden über das Verb WHEN gestellt, der JA-Zweig wird durch 

THEN definiert. Der NEIN-Zweig, bezogen auf alle vorausgegangenen WHEN-Klauseln, wird mit 

OTHERWISE kodiert. Sehen wir uns ein kleines Beispiel dazu an: 

SAY "Wähle unter folgenden Möglichkeiten" 

SAY;SAY 

SAY "1 <=> Black Jack" 

SAY "2 <=> Knobeln" 

SAY "3 <=> Mastermind" 

SAY "4 <=> Poker";SAY 

SAY "9 <=> Programmende" 

PULL auswahl 

SELECT 

   WHEN auswahl = 9 THEN EXIT 

   WHEN auswahl = 1 THEN CALL BLK_JACK 

   WHEN auswahl = 2 THEN CALL KNOBEL 

   WHEN auswahl = 3 THEN CALL MST_MIND 

   WHEN auswahl = 4 THEN CALL POKER 

   OTHERWISE SAY "Auswahl falsch" 

END 

Stellen wir dem SELECT-Gebilde eine IF-Konstruktion mit gleicher Auswirkung gegenüber, erübrigt 
sich jeglicher Kommentar: 

IF auswahl = 9 then exit 

ELSE IF auswahl = 1 THEN CALL BLK_JACK 

     ELSE 

        IF auswahl = 2 THEN CALL KNOBEL 

        ELSE 

           IF auswahl = 3 THEN CALL MST_MIND 

           ELSE 

              IF auswahl = 4 THEN CALL POKER 

              ELSE SAY "Auswahl falsch" 

Das SELECT-Gebilde lässt sich durch seine klare und einfache Struktur wesentlich besser lesen, ist 
leichter verständlich und macht auch dem, der ein Programm nicht geschrieben hat, viel leichter 
verständlich, was es bezwecken soll. Lesbarkeit und Änderungsfreundlichkeit gewinnen deutlich. 
Sehen wir uns an dieser Stelle die Datumsüberprüfung in Form einer Mehrfachabfrage an: 

SAY "Gib Dein Geburtsdatum an (Format: tt.mm.jj)" 

PULL gebdat 

PARSE VAR gebdat 1 tt 3 x1 4 mm 6 x2 7 jj 

MAX02 = 28+(jj//4=0) 

SELECT 

   WHEN LENGTH(gebdat) <> 8       THEN SAY "Falsche Länge" 

   WHEN x1!!x2 <> ".."            THEN SAY "Trennzeichen falsch" 

   WHEN ^DATATYPE(tt!!mm!!jj,"W") THEN , 

      SAY "tt/mm/jj nicht numerisch" 

   WHEN tt < 01 ! tt > 31         THEN SAY "Tag > 31" 

   WHEN mm < 01 ! mm > 12         THEN SAY "Monat falsch" 

   WHEN POS(mm,"04 06 09 11") > 0 & tt > 30 THEN , 

      SAY "Monat" mm "hat nur 30 Tage" 

   WHEN mm = 02 & tt > MAX02      THEN SAY "Februar hat" MAX "Tage" 

   OTHERWISE NOP 

END 

 

SAY "Gib Dein Geburtsdatum an (Format: tt.mm.jj)" 

PULL gebdat 

PARSE VAR gebdat 1 tt 3 x1 4 mm 6 x2 7 jj 

TIM = "31" 28+ (jj//4=0)-(jj//100=0)+(jj//400=0) , 

      "31 30 31 30 31 31 30 31 30 31" 

SELECT 

   WHEN LENGTH(gebdat) <> 8       THEN SAY "Falsche Länge" 

   WHEN x1!!x2 <> ".."            THEN SAY "Trennzeichen falsch" 

   WHEN ^DATATYPE(tt!!mm!!jj,"W") THEN SAY "tt/mm/jj nicht numerisch" 

   WHEN mm < 01 ! mm > 12         THEN SAY "Monat falsch" 

   WHEN tt > word(TIM,mm)         THEN SAY "Monat" mm "hat nur" , 

                                           word(TIM,mm) "Tage" 

   OTHERWISE NOP 

END 

1.21. Schleifenverarbeitung 

Schleifen sind immer angesagt, wenn Programmbefehle wiederholt auszuführen sind. Wenn wir da-
von ausgehen, dass die SIGNAL-Anweisung nicht als GOTO-Ersatz missbraucht wird (wer es tut, 
hat verdient, was er bekommt!), wird die Schleife auch in Verbindung mit Plausibilitätsprüfungen von 
Eingaben genutzt. Wird eine Eingabe als falsch erkannt, soll eine Fehlermeldung ausgegeben und 

dem Bediener eine neue Eingabe erlaubt werden. Die Rückkehr an einen Punkt des Programmes 
nach oben kann nur über eine Schleife erfolgen (der Maximalpunkt für die Rückkehr nach oben ist 
der Schleifenkopf einer aktiven Schleife). REXX bietet fünf verschiedene Formen der Schleife an. 
Dies wirft oft die Frage auf, welche Schleife für eine Problemstellung benutzt werden soll. Prinzipiell 
kann jedes Problem mit jeder der fünf Möglichkeiten gelöst werden, doch zeigt die praktische An-
wendung, dass sich meist eine Form geradezu aufdrängt. Gleichgültig, für welche Form man sich 
entscheidet, eines haben alle Varianten gemeinsam: Der Beginn der Schleife (Schleifenkopf) wird 
immer mit DO eingeleitet, das Schleifenende wird mit END markiert. Zwischen Schleifenbeginn und 

-ende werden die Befehle kodiert, die wiederholbar sein sollen. Das Statement DO steht nie alleine, 
sondern wird immer von der Schleifenbedingung gefolgt. 

1.21.1. Die simple Wiederholung (DO zähler) 

Hinter DO wird die Anzahl der Schleifendurchläufe angegeben. REXX vermindert die Zahl bei jedem 
Durchlauf um eins. Ist der Zähler bei 0 angelangt, wird hinter das zur Schleife gehörende END ver-
zweigt und die Verarbeitung fortgesetzt. Der Schleifenzähler muss Integer sein, kann aber auch über 
eine Variable geliefert werden. 

SAY CENTER("< Zentrierte Titelzeile >",79,"-") 

DO 3 

   SAY 

END 

Der SAY-Befehl in der Schleife wird dreimal ausgeführt und erzeugt nach der Überschriftszeile drei 
Leerzeilen. 

Um ein weiteres praktisches Beispiel für die einfache Schleife zu zeigen, nehmen wir eine Anleihe 
bei der Dateiverarbeitung, ohne im Detail darauf einzugehen (siehe Dateiverarbeitung im gleichen 
Kapitel oder EXECIO im Band-2 (Befehlssyntax)). 

"EXECIO * DISKR EING (FINIS" 

DO QUEUED() 

   PARSE PULL satz 

   SAY satz 

END 

Gibt alle Datensätze, die durch EXECIO in den STACK-Bereich gelesen wurden, am Terminal aus. 

1.21.2. Die Zählschleife (DO variable=zahl TO endewert BY ± zahl) 

Sie findet immer Anwendung, wenn während der Schleifendurchläufe wichtig ist, zum wievielten Mal 
die Schleife durchlaufen wird, oder ein Zähler benötigt wird, der bei jedem Durchlauf der Schleife 
gleichmäßig nach oben oder nach unten verändert werden soll. Vorteil dieser Schleife ist, dass man 
sich um das Verändern des Zählers (wir sprechen hier auch von einer Schleifensteuervariablen) 
innerhalb der Schleife nicht zu kümmern braucht. Um die Vorteile zu veranschaulichen nehmen wir 
ein simples Beispiel und sehen uns die Lösung dazu zunächst in Form der einfachen Schleife an. 

Ein Anwender tippt am Terminal seinen Namen ein. Das Programm gibt den Namen gesperrt ge-
schrieben (zwischen je zwei Zeichen wird BLANK eingefügt) wieder aus: 

SAY "Gib Deinen Namen ein" 

PARSE EXTERNAL name 

count = 1 

neu_name = "" 

DO LENGTH(name) 

   neu_name=neu_name SUBSTR(name,count,1) 

   count = count + 1 

END 

neu_name = STRIP(neu_name,L) 

SAY neu_name 

Um bei jedem Durchlauf auf ein anderes Byte des Namens zugreifen zu können, müssen wir eine 
Variable zunächst mit 1 initialisieren (beim 1. Durchlauf wollen wir ja an das erste Byte), und diese 
Variable in der Schleife nach erfolgter Verarbeitung um  1 erhöhen. So gelangen wir beim zweiten 
Durchlauf an das zweite Byte des Namens und so weiter. Hier drängt sich die Zählschleife auf. Zum 
einen kann die Initialisierung der schleifensteuernden Variablen in die Schleife eingebunden werden, 
zum anderen müssen wir uns um die Erhöhung der Steuervariablen im Programm nicht kümmern: 

SAY "Gib Deinen Namen ein" 

PARSE EXTERNAL name 

neu_name = "" 

DO count = 1 TO LENGTH(name) 

   neu_name=neu_name SUBSTR(name,count,1) 

END 

neu_name = STRIP(neu_name,L) 

SAY neu_name 

Die Aussage, dass die Steuervariable bei jedem Durchlauf um eins erhöht wird, entspricht dem 
Standard und muss nicht kodiert werden. Ist die Schrittweite der Erhöhung nicht 1, oder soll von der 
Steuervariablen bei jedem Durchlauf ein bestimmter Wert subtrahiert werden, muss die Angabe er-
folgen. Sehen wir uns das gleiche Beispiel nochmals an und verarbeiten die Zeichenkette hier von 
rechts nach links: 

SAY "Gib Deinen Namen ein" 

PARSE EXTERNAL name 

neu_name = "" 

DO count = LENGTH(name) to 1 BY -1 

   neu_name=SUBSTR(name,count,1) neu_name 

END 

neu_name = STRIP(neu_name,T) 

SAY neu_name 

Im nächsten Beispiel wollen wir alle Zahlen zwischen 1000 und 2000, die ohne Rest durch 25 teilbar 
sind, am Terminal ausgeben: 

DO count = 1000 to 2000 BY +25 

   SAY count 

END 

Das gleiche Ergebnis wäre zu erzielen, wenn kodiert wird: 

DO count = 1000 TO 2000 

   IF count // 25 = 0 THEN SAY count 

END 

Für ein scheinbar gleiches Ergebnis muss die untere Schleife 1001 mal durchlaufen werden, die 
obere Schleife liefert ihr Ergebnis mit nur 41 Durchläufen. Außerdem erfolgen in der unteren Schleife 
auch noch 1001 Prüfungen, von deren Ergebnis abhängig eine Bildschirmausgabe erfolgt. Die un-
tere Schleife benötigt etwa die 6fache Zeit. 

1.21.3. Schleifensteuernde Befehle 

An dieser Stelle wollen wir uns mit zwei Befehlen auseinandersetzen, die nur innerhalb der Pro-
grammschleife sinnvoll einzusetzen sind: ITERATE und LEAVE. 

ITERATE führt zum einmaligen Schleifendurchlauf, ohne dass die Befehle in der Schleife, die nach 
ITERATE folgen, ausgeführt werden. Subjektiv gewinnt man den Eindruck, dass ITERATE direkt 
zum Ansprung des Schleifenkopfes führt. Der Befehl wird häufig in einer Schleife für Plausibilitäts-
prüfungen eingesetzt, wenn ein Eingabefehler erkannt wurde und weitere Befehle nicht ausgeführt 
werden sollen. Am Schleifenkopf wird dann in aller Regel eine Neueingabe des Bedieners ermög-
licht. 

LEAVE führt zum Absprung aus der aktiven Schleife. Obwohl LEAVE im weiteren Sinne auch ein 
Sprungbefehl ist, kann er nicht mit GOTO verglichen werden, da durch LEAVE nicht ein beliebiger 
Punkt im Programm aufgesucht werden kann. Es wird immer hinter das zur Schleife gehörende END 
verzweigt, dort wird die Verarbeitung fortgesetzt. Soll aufgrund einer Bedingung aus mehr als einer 
Schleife gesprungen werden, kann LEAVE nicht genutzt werden (in Ausnahmefällen kann hierzu 
SIGNAL kodiert werden, mehr dazu aber unter SIGNAL im Kapitel »Befehlssyntax«). 

1.21.4. Die Endlosschleife (DO FOREVER) 

Die Endlosschleife (DO FOREVER) zwingt einen zu kodieren, was man auf gar keinen Fall will. Sie 
ist aber in vielen Fällen eine sehr praktische Variante, für die Wiederholbarkeit von Befehlen zu 
sorgen, ohne sich komplizierte Bedingungen auszudenken, die das Ende der Schleife beschreiben. 
Drei Situationen werden in der Praxis sehr gerne über die FOREVER-Schleife geregelt: 

1. Ein Programm soll beliebig oft wiederholbar sein und durch Bediener-Eingabe beendet wer-
den. 

2. Datenerfassungsschleifen mit Terminaleingabe. 
3. Bedienereingabe mit Plausibilitätsprüfungen. 

Sehen wir uns zu diesen Situationen je ein kleines Beispiel an: 

/* REXX * GAMEBOX *******************************/ 

DO FOREVER 

   SAY CENTER("< Heinzi's Spielebox >",79,"-");SAY 

   SAY "Wähle durch die entsprechende Zahl";SAY 

   SAY "1 <=> Black Jack" 

   SAY "2 <=> Knobeln" 

   SAY "3 <=> Mastermind" 

   SAY "4 <=> Poker";SAY 

   SAY "9 <=> Programmende" 

   SAY;SAY;SAY 

   SAY "Eingabe ??" 

   PULL auswahl 

   SELECT 

      WHEN auswahl = 9 THEN LEAVE 

      WHEN auswahl = 1 THEN CALL BLK_JACK 

      WHEN auswahl = 2 THEN CALL KNOBEL 

      WHEN auswahl = 3 THEN CALL MST_MIND 

      WHEN auswahl = 4 THEN CALL POKER 

      OTHERWISE SAY "Falsche Auswahl, nochmal" 

   END 

END 

EXIT 

Das Ende des Programmes wird erreicht, wenn der Bediener sich für die Auswahl 9 (Programm-
ende) entscheidet. Grundsätzlich hätte als Aktion auf diese Eingabe hin das Programm auch unmit-
telbar beendet werden können (WHEN auswahl = 9 THEN EXIT), es ist aber besser nur ein 

ENDE im Programm zu definieren. LEAVE verzweigt hinter das END der aktiven Schleife (es gibt ja 
nur eine) und trifft dort auf den Befehl EXIT. Das Programm wird beendet. 

/* REXX * TERM_EIN */ 

cnt=0 

SAY "Eingabe (ENTER = Ende)" 

DO FOREVER 

   PARSE EXTERNAL ein 

   IF LENGTH(ein) = 0 THEN LEAVE 

   cnt = cnt + 1 

   eingabe.cnt = ein 

END 

SAY "Empfangene Eingaben:" cnt 

/* Hier könnten die Daten verarbeitet werden */ 

EXIT 

In diesem Beispiel kann der Anwender beliebig viele Eingaben am Terminal durchführen. Drückt er 
<Enter>, wird die Eingabeschleife beendet. Sofern in Verbindung mit <Enter> Daten eingegeben 
werden (auch BLANK sind Daten), wird Eingabe für Eingabe ein Variablenstamm aufgebaut. Die 
erste Eingabe liegt in der Variablen EINGABE.1, die zweite in der Variablen EINGABE.2 und so 
weiter. Wie viele gültige Eingaben in der Schleife erfolgten, kann aus dem Wert von CNT abgelesen 
werden. 

Im folgenden Beispiel gehen wir davon aus, dass der Bediener den Namen einer zu verarbeitenden 
Datei einzugeben hat. Der Name kann voll- oder teilqualifiziert eingegeben werden. Wird eine voll-
qualifizierte Eingabe gemacht, müssen die Apostrophe mit getippt werden. Die Datei soll über IEB-
GENER ausgedruckt werden. Folgende Problemsituationen werden abgefangen: Datei nicht verfüg-
bar, PO-Datei ohne Memberangabe, Satzlänge zu groß, Dateiformat ungleich PS und PO. 

/* REXX * DRUCK *********************************/ 

DO FOREVER 

   SAY "Name Druckdatei (ggf. mit Hochkomma)" 

   PARSE UPPER EXTERNAL drudat 

   IF LENGTH(drudat) = 0 THEN EXIT 

   IF SYSDSN(drudat) = "OK" THEN DO 

      DMY = LISTDSI(drudat) 

      LastByte = right(strip(drudat,,"'"),1) 

      SELECT 

         WHEN POS(SYSDSORG,"PS PO") = 0 THEN msg = "Falsche DS-Org" 

         WHEN SYSDSORG = "PO" & ,       /* PO-Datei ohne Memberangabe */ 

            LastByte >< ")"             THEN msg = "PO-Datei ohne Member" 

         WHEN SYSLRECL > 132            THEN msg = "Satzlänge zu groß" 

         OTHERWISE LEAVE 

      END 

   END 

   ELSE msg = drudat "nicht verfügbar" 

   SAY msg 

END 

"ALLOC DD(SYSIN) DUMMY" 

"ALLOC DD(SYSPRINT) DUMMY" 

"ALLOC DD(SYSUT1) DA("drudat") SHR REUSE" 

"ALLOC DD(SYSUT2) SYSOUT(X)" 

"IEBGENER" 

"FREE DD(SYSIN SYSPRINT SYSUT1 SYSUT2)" 

EXIT 

Die FOREVER-Schleife kann aus zwei Gründen verlassen werden. Der Anwender drückt auf die 
Frage nach der Druckdatei nur <ENTER> ohne Dateneingabe (in diesem Fall wird das Programm 
unmittelbar beendet. Einen Ausstieg in dieser oder ähnlicher Form sollte man immer schaffen, für 
den Fall, dass der Anwender eigentlich ein anderes Programm starten wollte), oder der eingegebene 
Dateiname spricht nicht gegen eine Verarbeitung. In diesem Fall ist sichergestellt: 

▪ RACF hat keine Einwände 
▪ Das Dateiformat ist OK. 
▪ Wurde eine PO-Datei angegeben, ist auch eine Memberangabe erfolgt. 
▪ Die Satzlänge übersteigt 132 Byte nicht. 

Treffen alle diese Dinge zu, befinden wir uns innerhalb der Mehrfachabfrage im NEIN-Zweig 
(OTHERWISE) und verlassen die Schleife nach unten, wo die Zuweisungen für den IEBGENER und 
dessen Aufruf stattfinden. 

Die letzten beiden Formen der Programmschleife sehen auf den ersten Blick gleich aus. Scheinbar 
besteht der Unterschied nur darin, dass die eine Schleife beschreibt, wie lange eine Verarbeitung 
wiederholt werden soll, die andere dagegen festlegt, wann die Wiederholung als beendet betrachtet 
wird. Kehrt man die schleifensteuernde Bedingung ins Gegenteil, erreicht man scheinbar Gleiches. 
Von da her gesehen hätte man auf eine der beiden Formen verzichten können (unter CLIST war die 
UNTIL-Schleife vor TSO Release 2 nicht möglich). Bei Licht betrachtet gibt es aber einen ganz gra-
vierenden Unterschied, der im Zeitpunkt der Überprüfung der schleifensteuernden Bedingung liegt. 

Die WHILE-Bedingung wird am Schleifenanfang geprüft. Trifft die Bedingung zu, wird die Schleife 
betreten. Anderenfalls wird hinter das zur Schleife gehörende END verzweigt. Wir sprechen von 
einer PRE-abweisenden Schleife. 

Die UNTIL-Schleife wird zunächst in jedem Fall einmal durchlaufen. Die schleifensteuernde Bedin-
gung wird am Schleifenende überprüft. Ist die Bedingung erfüllt, wird die Schleife verlassen, ande-
renfalls wird sie aufs Neue durchlaufen. In diesem Fall sprechen wir von einer POST-abweisenden 
Schleife. 

Beide Schleifen finden zu ihrem geregelten Ende, sobald eine Bedingung wahr (UNTIL) oder unwahr 
(WHILE) ist. Unabhängig davon können beide Schleifen aber auch über LEAVE verlassen werden. 
Am besten veranschaulichen wir die beiden Schleifenvarianten mit ein paar Beispielen, die verdeut-
lichen, in welchen Situationen die jeweilige Schleife sinnvoll eingesetzt werden kann, wo unter Um-
ständen beide Schleifen zum Erfolg führen können, oder nur eine der beiden Varianten die sinnvol-
lere ist. 

1.21.5. Die PRE-abweisende Schleife (DO WHILE bedingung) 

Gehen wir von folgender Situation aus: Ein Programm benötigt für die Verarbeitung eine Zahl vom 
Anwender. Diese Zahl sollte bei Programmstart bereits mitgeliefert werden können. Wurde bei Aufruf 
des Programmes keine Zahl übergeben, muss das Programm nachfragen. 

/* REXX * PZ  

*/ 

ARG zahl 

DO WHILE DATATYPE(zahl) = "CHAR" 

   SAY "Gib eine Zahl ein" 

   PARSE EXTERNAL zahl 

END 

/* Hier liegt die Verarbeitung */ 

EXIT 

Hier macht die UNTIL-Schleife keinen Sinn. Sie würde auch dann einmal durchlaufen werden, wenn 
die benötigte Zahl durch die Instruktion ARGument bereits zugewiesen worden wäre. 

Wir können die WHILE-Schleife auch für vereinfachte Plausibilitätsprüfung benutzen. Soll eine Datei 
im Programm über den ALLOCATE-Befehl erzeugt werden und der Bediener unter anderem die 
Dateiorganisationsform, bei der wir nur sequentiell oder gegliedert zulassen wollen, eingeben, 
könnte das folgendermaßen aussehen: 

DO WHILE dsorg >< "PS" & dsorg >< "PO" 

   SAY "DS-Organisationsform eingeben (PS oder PO)" 

   PULL dsorg 

END 

Hier kommt uns das Variablenhandling von REXX sehr gelegen. Sofern die Variable DSORG bis 
dahin noch nie benutzt wurde, hätten wir in den meisten Sprachen an dieser Stelle Programmab-
bruch. In REXX dagegen zeigt die Variable auf ihren zu Großschreibung konvertierten Eigennamen, 
also "DSORG". Aus diesem Grund wäre die Bedingung am Schleifenkopf (Inhalt der Variablen 
DSORG ist weder "PS" noch "PO") erfüllt, die Schleife wird betreten und so lange durchlaufen, bis 
der Bediener die Variable mit der Zeichenfolge "PS" oder "PO" füllt. 

1.21.6. Die POST-abweisende Schleife (DO UNTIL bedingung) 

Nehmen wir an, ein Bediener soll in einer Schleife durch Eingabe entscheiden, ob eine bestimmte 
Aktion durchgeführt werden soll oder nicht. In vielen Fällen wird hier eine Ja/Nein Eingabe am Ter-
minal ermöglicht. Häufig wird einer der beiden Fälle (JA oder NEIN) geprüft und behandelt. Alle 
anderen möglichen Eingaben werden gleich (negativ) behandelt. 

/* REXX * DEMO */  

DO FOREVER 

   /* Programmverarbeitung */ 

   SAY "Noch ein Durchlauf (J/N)" 

   PULL antwort 

   IF antwort = "N" THEN LEAVE 

END 

EXIT 

Ein gängiges Verfahren. Wie sinnvoll es ist, möchte ich hier nicht kommentieren. Was dabei heraus-
kommen kann, zeigt am besten das folgende kleine Beispiel 

/* REXX * DEMO */ 

CALL OUTTRAP "datei." 

"LISTC" 

CALL OUTTRAP "OFF" 

DO cnt = 2 TO datei.0 

   SAY datei.cnt "löschen oder behalten (L/B)" 

   PULL antwort 

   IF antwort = "B" THEN , 

      SAY datei.cnt "bleibt erhalten" 

   ELSE DO 

      "DEL '"datei.cnt"'" 

      SAY datei.cnt "gelöscht" 

   END 

END 

EXIT 

Wird auf die Frage nach dem Löschen mit B geantwortet, bleibt die Datei erhalten. Für alle anderen 
Fälle, auch Tippfehler, erfolgt das Löschen. Prima! Sinnvoller wäre hier im Programm bereits sicher-
zustellen, dass die Bedienereingabe nur die Werte liefern kann, die das Programm erwartet: 

DO UNTIL antwort = "L" ! antwort = "B" 

   SAY datei.cnt "löschen oder behalten (L/B)" 

   PULL antwort 

END 

"HALT" wird der aufmerksame Leser hier einwenden. Es geht genauso gut mit: 

DO WHILE antwort >< "L" & antwort >< "B" 

   SAY datei.cnt "löschen oder behalten (L/B)" 

   PULL antwort 

END 

was bedingt richtig ist. Da die Abfrageschleife in unserem Beispiel aber in eine äußere Programm-
schleife für die Abarbeitung der einzelnen Dateinamen integriert ist, tritt ein Problem auf: Wird die 
WHILE-Schleife zum ersten Mal angegangen, ist der Inhalt der Variablen ANTWORT ihr großge-
schriebener Eigenname und erfüllt damit die Bedingung ungleich L und ungleich B. Die Schleife wird 
betreten, der Anwender kann seine Eingabe machen. Gelingt es ihm, ein L oder B zu tippen, wird 
die Schleife verlassen und die entsprechende Aktion (Datei löschen oder nicht löschen) wird durch-
geführt. Das Programm kehrt an den äußeren Schleifenkopf zurück. Steht der Interpreter erneut am 
Kopf der WHILE-Schleife, ist ihre Bedingung nicht erfüllt (ANTWORT enthält immer noch das L oder 
B der ersten Eingabe). Die Schleife wird übersprungen. Auch für die zweite und alle weiteren Dateien 
erfolgt die gleiche Aktion wie für die erste. Im Falle Löschen der ersten Datei wird dies dem Anwen-
der sicher Tränen in die Augen treiben. Andererseits schafft es natürlich enormen Platz auf dem 
Plattenpool! 

Damit die WHILE-Schleife hier vernünftig funktionieren kann, muss die Variable ANTWORT vor er-
neutem Anlauf des Schleifenkopfes wieder zurückgesetzt werden. Theoretisch könnte der Befehl 
dazu an beliebiger Stelle innerhalb der äußeren Schleife liegen. Im Hinblick auf spätere Program-
mänderungen ist es am sinnvollsten, das Rücksetzen der Variablen direkt über dem Schleifenkopf 
durchzuführen. An dieser Stelle erklärt sich der Befehl selbst. 

DROP antwort 

DO WHILE antwort >< "L" & antwort >< "B" 

   SAY datei.cnt "löschen oder behalten (L/B)" 

   PULL antwort 

END 

Wird dagegen die UNTIL-Schleife eingesetzt, wird sie zunächst einmal sicher durchlaufen, gleich-
gültig, ob beim ersten oder einem folgenden Durchlauf der äußeren Schleife. Der Anwender be-
kommt so in jedem Fall bei jedem Schleifendurchlauf die Möglichkeit, seine Antwort auf die Frage 
nach der Aktion einzugeben. Ein Rücksetzen der Variablen ANTWORT ist nicht nötig. 

An dieser Stelle möchte ich eine eindringliche Bitte an Euch richten: Verzichtet auf jeden Befehl, den 
ich nicht unbedingt benötigt. Alles, was kodiert wird, muss stimmen. Nur die Befehle, die nicht ge-
schrieben werden, können nicht falsch sein. Der eine oder andere Befehl mehr ist nur sinnvoll, wenn 
die Lesbarkeit des Programmes dadurch verbessert wird, wobei dies ein relativer Begriff ist, abhän-
gig vom Kenntnisstand des Programmierers. 

1.21.7. Kombinationen 

Achtung, wir bewegen uns auf eine gefährliche Stelle zu. Grundsätzlich können die verschiedenen 
Schleifen auch miteinander kombiniert werden. Wenn Ihr Seelenheil als Programmierer nicht davon 
abhängt, versuchet es zu vermeiden. Nicht alles, was theoretisch machbar ist, ergibt in der Praxis 
einen Sinn. Ein Schleifenkopf wie 

DO §=1 TO 1E6 WHILE z < 98 & do_it = "J" UNTIL x > y 

ist, was den reinen Formalismus angeht, ohne weiteres möglich. Ich bedauere aber bei derartigen 
Konstrukten immer den armen Mitarbeiter, der ein so schlimmes Gebilde im Nachhinein zu pflegen 
oder zu ändern hat. Leider sind der Erzeuger und der Änderer eines Programmes in aller Regel 
verschiedene Personen. Es ist sehr aufwendig, sich in eine Programmlogik mit komplexen und kom-
plizierten Verknüpfungen und Bedingungen einzulesen, wenn man sie nicht selbst geschrieben hat. 
Und selbst dann fällt es einem oft schwer, nachzuvollziehen, was man vor Jahren gedacht hat, als 
man derartige, monströse Gebilde ersonnen hatte. REXX-Programme werden häufig nicht oder nur 
sehr spärlich dokumentiert. Die Kunst der Programmierung besteht nicht darin, möglichst kompli-
zierte Programme zu schreiben, sondern einen Ablauf in simple, leicht verständliche Schritte zu 
zerlegen und in möglichst einfache und leicht lesbare Befehle umzuwandeln. Kompliziert wird ein 
Programm oft von ganz alleine. Programmierer, die durch eine gewollt komplizierte Kodierung ihren 
Arbeitsplatz sichern wollen, sollten heute eigentlich der Vergangenheit angehören. 

1.21.8. Keine Regel ohne Ausnahme 

Es gibt Situationen, in denen eine Vermengung unterschiedlicher Schleifenkonstruktionen von gro-
ßem Vorteil sein kann. Wir sehen uns mal eine an. Gehen wir von folgender Situation aus: Ein Pro-
gramm benötigt vom Anwender Informationen direkt bei Aufruf. Natürlich sollte der Anwender dar-
über informiert sein, doch oftmals sieht die Praxis eben ein wenig anders aus als die Theorie. Auf 
PC-Ebene hat sich durchgesetzt, dass ein Programm durch Startparameter wie /H oder /? eine 
kleine Hilfestellung über seine Bedienung liefert. Vergleichbares wollen wir auch in unsere REXX-
Programme integrieren. Ausgehend davon, dass ein Programmkopf kodiert wird, in dem die wich-
tigsten Informationen über die Bedienung eines Programmes stehen, wollen wir erreichen, dass, 
wenn ein Bediener keine oder falsche Angaben beim Startaufruf übergibt, oder über Startaufruf ein 
Fragezeichen(?) geliefert wird, dieser Programmkopf als Hilfestellung für den Bediener ausgegeben 
wird. In unserem Beispiel erfolgt die Ausgabe des Schleifenkopfes ab der Zeile, in der die Zeichen-
folge "Start" vorkommt. 

/* REXX * LOTTO ********************************/ 

/* Erstellt: 04.2009 Wittemann                 */ 

/*                                             */ 

/* parm1..: Gesamtanzahl Ziehungszahlen        */ 

/* parm2..: Anzahl zu ziehender Zahlen         */ 

ARG parm1 parm2 test . 

PARSE SOURCE . . progname . 

IF DATATYPE(parm1) = "CHAR" THEN DO 

   SAY "Online-Hilfe zu Programm" progname 

   s_ausg = 0 

   DO §=1 WHILE LEFT(SOURCELINE(§),2) = "/*" 

      IF POS("Start",SOURCELINE(§)) > 0 THEN s_ausg=1 

      IF s_ausg = 1 THEN SAY SOURCELINE(§) 

   END 

   EXIT 

END 

/* Hier liegt die eigentliche Verarbeitung */ 

EXIT 

Die ARG-Instruktion kann drei Variable füllen. Die ersten zwei werden im Programm zwingend als 
numerische Werte benötigt. Der Name des laufenden Programmes wird  in PROGNAME hinterlegt. 
Ist der erste Übergabewert nicht numerisch (ist auch erfüllt, wenn keine Wertübergabe stattgefunden 
hat), soll eine ONLINE-Hilfe ausgegeben werden. Hierzu eine aus zwei Varianten zusammenge-
setzte Schleifenbedingung. 

Der Ansatz §=1 führt durch das fehlende TO wert BY ± zahl zu einer endlosen Zählschleife, 

bei der §  mit 1 initialisiert und bei jedem Durchlauf um 1 erhöht wird. Über diesen Zeiger wird der 
Quellencode des laufenden Programmes ab der ersten Zeile gelesen. Das Ende der Schleife wird 
über WHILE gesteuert und ist erreicht, wenn eine Zeile auf Stelle 1 und 2 nicht /* führt. Auf diese 

Weise wird der gesamte Programmkopf verarbeitet. In der Schleife suchen wir nach dem Wort "Start" 
(aus Startaufruf). Ist es gefunden, wird ein Schalter für die Ausgabe des Quellcodes aktiviert. Diese 
und alle weiteren Zeilen des Programmkopfes werden am Terminal gezeigt. Dieses Verfahren ist 
sicher etwas aufwendiger, als auf das Beispiel bezogen Zeile 11 bis 17 auszugeben, aber es pflegt 
sich quasi selbst mit, wenn im Programmkopf Änderungen durchgeführt werden. 

1.21.9. Plausibilitätsprüfungen 

Immer, wenn der Bediener in den Ablauf eines Programmes integriert werden muss, wird es gefähr-
lich. Sitzt man vor einem Terminal (Anwender, oder Neudeutsch PEBCAK = Problem Exists Between 
Chair And Keyboard), ist EDV in aller Regel nur Mittel zum Zweck. Meist will man EDV nur benutzen, 
ohne Wert darauf zu legen, wie und warum etwas funktioniert. Steht man dagegen hinter dem Ter-
minal (als verantwortlicher Programmierer oder Organisator einer Anwendung), sollte man sicher-
stellen, dass ein Programm auch mit widrigen Umständen zurechtkommt. Sehr früh muss man sich 
die Frage stellen: "Was kann der Bediener alles falsch machen?". Die Antwort lautet immer: "AL-
LES!!" Will man bei Mitwirkung des Bedieners nicht alles hinnehmen, was dieser über die Tastatur 
zum Ablauf eines Programmes beiträgt, sind Prüfungen auf formale Richtigkeit der Eingaben unver-
zichtbar. Diese Prüfungen werden meist als einfache Abfragen oder Mehrfachabfragen durchge-
führt. Werden Fehler bei der Eingabe erkannt, könnte das Programm die betreffenden Variablen auf 
Standardwerte setzen, oder für den Bediener eine Fehlermeldung ausgeben und ihm eine Neuein-
gabe ermöglichen. Im zweiten Fall muss über eine Schleife der Part der Bedienereingabe wieder-
holbar gemacht werden. 

Im Anschluss sehen wir uns die gängigsten Funktionen an (REXX-intern und erweiterte TSO-Funk-
tionen), die in Verbindung mit Plausibilitätsprüfungen eingesetzt werden, und ihre Bedeutung: 

Funktion Wertrückgabe, Bedeutung 

DATATYPE(var) NUM oder CHAR, lässt den Datentyp einer Variablen erkennen (nume-
risch oder character). 

DATATYPE(var,typ) 0(nein) oder 1(ja), teilt mit, ob eine Variable dem Datentyp typ entspricht. 

INDEX(var1,var2) Numerischer Wert, welcher der Byteposition von var2 innerhalb var1 ent-
spricht, an der var2 gefunden wurde. Ist var2 in var1 nicht enthalten, wird 
0 zurückgegeben. 

LEFT(var,anzahl) Übernimmt von links eine bestimmte Anzahl Byte. 

LISTDSI(datei) Liefert viele Informationen über eine Datei in dafür vorgesehenen Variab-
len 

POS(var1,var2) Numerischer Wert, welcher der Byteposition von var1 innerhalb var2 ent-
spricht. Ist var1 in var2 nicht enthalten, wird 0 zurückgegeben. 

RIGHT(var,anzahl) Übernimmt aus einer Variablen von rechts her eine bestimmte Anzahl 
Byte. 

SUBSTR(var,pos,an-
zahl) 

Übergibt eine bestimmte Anzahl Byte ab der angegebenen Position aus 
einer Variablen. 

SYSDSN(datei) Lässt erkennen, ob eine Datei verfügbar ist oder nicht. 

USERID() Liefert die Benutzerkennung des Aufrufers aus LOGON. 

1.22. Der STACK-Bereich 

Der STACK ist ein Bereich im TSO-Adressraum, der dynamisch verwaltet wird. Dieser Bereich kann 
auf unterschiedliche Weise genutzt werden. Zum einen können Datensätze durch REXX-Instruktio-
nen im Stack hinterlegt werden, zum anderen können Datensätze als Ergebnis einer Datei-Lese-
operation den Stack füllen. 

Sollen Sätze aus dem Stack verarbeitet werden, kann dies ebenfalls auf unterschiedliche Weise 
erfolgen. Über eine REXX-Instruktion kann ein Satz aus dem Stack entnommen und in einer Vari-
ablen abgelegt werden. Es gibt auch TSO-REXX-Befehle, die zum Entleeren des Stacks führen. 

Wird eine Schreibaktion auf Datei durchgeführt, bei welcher der Stackbereich als Input definiert ist, 
verlassen die Datensätze den Stack. Es findet hier also kein reiner Lesevorgang, sondern eine Über-
tragung statt. Nach Ausführung einer Schreibverarbeitung, die den gesamten Stack als Eingabe 
vorsieht, ist der Stackbereich leer. 

Sind im Stack Job-Control-Statements hinterlegt, kann durch den SUBMIT-Befehl in einer bestimm-
ten Form eine direkte Übergabe der Daten im Stack an die SUBMIT-Schnittstelle erfolgen. Auch 
hierbei wird der Stack entleert. Diese Form der Verarbeitung erlaubt, Job-Control-Gerippe in einer 
Datei zu hinterlegen, in der REXX-Variablen Bestandteil der Job-Control sind. Werden die State-
ments aus der Datei gelesen und im Stack abgelegt, werden die Variablen vor Übergabe an die 
Submit-Schnittstelle vom Interpreter durch ihre Werte ersetzt. Auf diese Weise kann ein Komfortgrad 
ähnlich zu File-Tailoring des Dialog Managers erreicht werden (ich bitte alle Dialog Manager Freaks 
für diese denkwürdige Aussage hier um Entschuldigung). 

Soll zwischen zwei Programmen Informationsaustausch im größeren Umfang in beiden Richtungen 
stattfinden, sind die Möglichkeiten durch CALL prog parameter - ARG variable - RETURN wert - 
Übernahme über RESULT oft sehr mühsam, in extremen Situationen bisweilen unmöglich. Auch 
hier kann der Stack als Zwischenträger der auszutauschenden Informationen genutzt werden. Na-
türlich wird der erfahrene TSO-Profi an dieser Stelle lieber auf die Variablenservices des Dialog 
Managers zurückgreifen, aber die wollen wir hier vorerst außer Acht lassen. 

Die Leerung des Stacks findet bei Ende des hierarchisch obersten REXX-Programmes in jedem Fall 
statt, auch wenn programmseitig keine Befehle kodiert sind. Bei Programmende werden die verblei-
benden Daten im Stack zum REXX-Environment (im allgemeinen TSO) hin übergeben. Dies eröffnet 
ebenfalls nicht zu unterschätzende Möglichkeiten. 

Generell kann der Stack auch untergliedert werden. Wird ein Puffer in den Stack eingebaut, sind nur 
die Daten des aktiven Pufferbereiches erreichbar. Leider ist eine Positionierung zwischen den ver-
schiedenen Pufferbereichen nicht möglich. An die Daten des vorgelagerten Puffers gelangt man nur, 
indem der aktive Pufferbereich aufgelöst wird. Enthält er zu diesem Zeitpunkt noch Daten, werden 
diese dem Verfall preisgegeben. 

Wird der Stackbereich gepuffert, verlangt dies nach einer grundliegend anderen Verarbeitungslogik. 
Informationen, welche die Anzahl der Datensätze im Stack liefern, sind wertlos. Es muss entspre-
chend der jeweilige Pufferbereich untersucht werden. Die einzelnen Pufferbereiche müssen sinn-
vollerweise so, wie sie erzeugt wurden, verarbeitet und wieder aufgelöst werden. Existieren bei-
spielsweise fünf Pufferbereiche, und ein Programm löscht den Puffer 2, werden die Pufferbereiche 
3 bis 5 ebenfalls aufgelöst. Die eventuell darin enthaltenen Daten sind verloren. Ich habe lange 
vergeblich nach einer vernünftigen Anwendung für die Gliederung von Stackbereichen gesucht. Alle 
meine bisherigen Probleme ließen sich auf andere Weise viel besser lösen. 

Wie bereits erwähnt, müssen wir uns um das Vorhandensein des Stackbereiches nicht kümmern. 
Grundsätzlich können beliebig viele derartige Bereiche aus einem REXX-Programm heraus erzeugt 
werden. Die einzelnen Stackbereiche werden dabei nummeriert, wobei der originäre Bereich immer 
die Nummer 1 trägt. Alle weiteren erzeugten Stacks werden entsprechend ihrer Erstellung um 1 
erhöht. Jeder einzelne kann sich bis 16MB ausdehnen. Die Anzahl der Stacks und deren Größe wird 
letztlich nur durch die Adressraumgröße der LOGON-Session begrenzt. Der Einsatz mehrerer 
Stacks kann in der Praxis in Ausnahmefällen ganz nützlich sein, erfordert aber auch große Umsicht 
bei der Programmierung. Wir werden uns hierzu noch ein Beispiel anschauen. Sind mehrere Stack-
bereiche erzeugt worden, kann kein wahlfreier Zugriff auf einen beliebigen dieser Bereiche erfolgen. 
Es ist immer nur der letzte erzeugte Stackbereich verfügbar (er überlagert alle anderen Bereiche). 
Um an einen vorgelagerten Stackbereich zu gelangen, muss der aktuelle Bereich gelöscht werden. 
Enthält der Bereich zum Zeitpunkt des Löschens noch Daten, werden diese verworfen. Wird der 
Löschbefehl auf den originären Stack angewandt, wird der Bereich in diesem Sinne nicht aufgelöst, 
aber alle darin enthaltenen Daten (und Pufferbereiche) sind verloren. 

Im Folgenden sehen wir uns einige Befehle und Funktionen an, die uns das Arbeiten mit dem Stack 
ermöglichen: 

 

Der Stack kann über zwei REXX-Instruktionen gefüllt werden. QUEUE legt immer einen neuen Da-
tensatz in den Stack und füllt ihn von unten nach oben. Dies entspricht dem Verarbeitungsprinzip F-
I-F-O (First In First Out). PUSH dagegen schiebt zuunterst im Stack einen Datensatz ein. Diese 
Form entspricht dem Prinzip L-I-F-O (Last In First Out), da bei der Entnahme eines  Datensatzes der 
unterste Datensatz den Stack als erster verlässt. 

Die Funktion QUEUED() liefert die aktuelle Anzahl der Datensätze im Stack. Die Schleife wird dem-
entsprechend viermal durchlaufen. Bei jedem Durchlauf wird der unterste Datensatz aus dem Stack 
entnommen, am Terminal angezeigt und anschließend von oben her wieder in den Stack gelegt. 
Stackbezogen haben wir nach Verarbeitung den gleichen Zustand wie vor Verarbeitung. 

/* REXX * DEMO  

*/ 

IF SYSVAR(SYSISPF) = "ACTIVE" THEN DO 

   SAY "Programm darf nicht unter ISPF/PDF laufen" 

   EXIT 

END 

QUEUE "LOGOFF" 

/* Verarbeitung */ 

EXIT 

Der TSO-Befehl LOGOFF wird im Stack hinterlegt, wenn sichergestellt ist, dass das Programm nicht 
unter PDF gestartet wurde (LOGOFF wird dort nicht unterstützt). Bei Programmende wird der Sta-
ckinhalt zum TSO übergeben, die TSO-Session wird automatisch abgebaut. 

DO c_loop = 1 to 3 

   "MAKEBUF" 

   c_rec = 0 

   DO FOREVER 

      SAY "Durchlauf" c_loop "Eingabe (ENTER = Ende)" 

      PARSE EXTERNAL ein 

      IF LENGTH(ein) = 0 THEN LEAVE 

      c_rec = c_rec + 1 

      queue ein 

   END 

   SAY "Eingaben im "c_loop". Durchlauf:" c_rec 

END 

Hier wird es nun etwas diffiziler. Wir haben zwei ineinander liegende Schleifen. Die äußere erlaubt 
dreimal eine beliebige Anzahl Eingaben (werden in der inneren Schleife ermöglicht). Die Erfassung 
endet, wenn der Bediener in einem Durchlauf der inneren Schleife <ENTER> drückt, ohne Daten 
einzugeben. 

Die einzelnen Datenpakete werden im Stack getrennt voneinander hinterlegt. Bereits das erste er-
zeugte Datenpaket wird in einem eigenen Pufferbereich abgelegt und dadurch von eventuell im 
Stack bereits vorhandenen Daten abgeschottet. Der Stack zeigt sich nach Verarbeitung folgender-
maßen: 

Wenn wir diese drei Datenpakete jetzt wieder verarbeiten wollen, können wir nicht die simple 
Schleife DO QUEUED() benutzen. Zum einen liefert QUEUED() nach wie vor die Gesamtzahl der 
Datensätze im Stack, zum anderen können wir nur die Datensätze verarbeiten, die sich im aktuellen 
Pufferbereich befinden. Um an die Datensätze des vorgelagerten Puffers zu gelangen, muss der 
aktuelle Puffer aufgelöst werden. Die Ausgabe der einzelnen Datenpakete könnte wie folgt ausse-
hen: 

SAY "Datenausgabe" 

DO c_loop = 3 to 1 BY -1 

   "QELEM" 

   say "Datenpaket" c_loop "("rc "Datensätze)" 

   DO RC 

      PARSE PULL satz 

      SAY satz 

   END 

   "DROPBUF" 

END 

Die Anzahl der Sätze im aktuellen Pufferbereich kann über den TSO-REXX-Befehl QELEM ermittelt 
werden. QELEM füllt die Variable RC mit der Anzahl der Datensätze im aktuellen Pufferbereich 
(queued elements). Wurden die Datensätze des aktuellen Pufferbereiches über die innere Schleife 
aus dem Stack entnommen, kann der Puffer aufgelöst (DROPBUF) und das vorgelagerte Datenpa-
ket freigegeben werden. 

So weit, so gut. Sinnvollerweise muss man davon ausgehen, dass die Anzahl zu hinterlegender 
Datenpakete nicht starr sein muss, und dass vor Beginn der Erfassung u.U. einige Puffer im Stack 
existieren. In diesem Fall muss die äußere Programmschleife der Erfassung (je Durchlauf ein Paket) 
flexibel sein. Dies führt zu einem anderen Problem. Es dürfen nur so viele Pufferbereiche verarbeitet 
werden, als im Erfassungsteil erzeugt wurden. Das Leben ist hart! 

"QBUF" 

anz_buf=rc 

DO c_loop=1 

   "MAKEBUF" 

   c_rec = 0 

   DO FOREVER 

      SAY "Durchlauf" c_loop "Eingabe (ENTER = Ende)" 

      PARSE EXTERNAL ein 

      IF LENGTH(ein) = 0 THEN LEAVE 

      c_rec = c_rec + 1 

      queue ein 

   END 

   SAY "Eingaben im "c_loop". Durchlauf:" c_rec 

   DO UNTIL ant = "J" ! ant = "N" 

      SAY "Neues Datenpaket erfassen (J/N)" 

      PARSE UPPER EXTERNAL ant 

   END 

   IF ant = "N" THEN LEAVE 

END 

: 

SAY "Datenausgabe" 

"QBUF" 

DO c_loop = rc - anz_buf to 1 BY -1 

   "QELEM" 

   say "Datenpaket" c_loop "("rc "Datensätze)" 

   DO RC 

      PARSE PULL satz 

      SAY satz 

   END 

   "DROPBUF" 

END 

Vor Beginn der Datenerfassung müssen wir festhalten, wie viele Pufferbereiche eventuell bereits 
existieren. QBUF hinterlegt die Anzahl der Puffer in der Variablen RC. Um den Wert nicht zu verlie-
ren kopieren wir ihn in die Variable ANZ_BUF. Die äußere Erfassungsschleife kann beliebig oft 
durchlaufen werden, bis der Anwender am Schleifenende keine weitere Verarbeitung mehr wünscht. 
Vor Ausgabe wird die Anzahl Puffer neu ermittelt. Die Anzahl durch die Erfassungsschleife erzeugter 
Puffer ergibt sich aus der Anzahl Puffer aktuell minus Anzahl Puffer vor Start der Erfassung. 

Im nächsten Beispiel stellen wir uns folgende Situation vor: Ein Programm hält Daten im Stack und 
muss ein Unterprogramm aufrufen. Natürlich sollten sich zwei Programmierer, die an einem Projekt 
arbeiten, mehr zu sagen haben als "Guten Appetit" beim Frühstück. Betrachtet man aber die Reali-
tät, erfüllt es mich immer mit grenzenlosem Staunen, dass Engländer und Franzosen unter dem 
Ärmelkanal nicht aneinander vorbeigegraben und sich in Australien getroffen haben. Weil man ja nie 
so recht weiß, was in einem untergeordnet aufgerufenen Programm so alles passiert, ist es sinnvoll, 
den aktuellen Stack zu sichern (vorausgesetzt, die Daten, die im Stack liegen, sollen nicht vom Un-
terprogramm verarbeitet werden). Und weil man oft nicht weiß, ob das Unterprogramm seinerseits 
wieder eines aufruft, oder das eigene Programm bereits UPRO eines anderen ist, hier ein Verfahren, 
den aktuellen Stack zu sichern und nach erfolgtem Unterprogramm zum richtigen Stack zurückzu-
finden, selbst wenn ein untergeordnetes Programm einen weiteren Stack erzeugt, aber nicht wieder 
gelöscht hat: 

"QSTACK" 

akt_stak = rc 

"NEWSTACK" 

CALL UPROG1 

"QSTACK" 

DO RC - akt_stak 

   "DELSTACK" 

END 

Die Instruktion QSTACK hinterlegt in der Variablen RC die Anzahl vorhandener Stackbereiche. Vor 
Aufruf des Unterprogrammes wird ein neuer Stackbereich erzeugt, der den alten Stack überlagert. 

Alle stackspezifischen Aktionen finden in diesem neuen Bereich statt. Nach Ende des Unterpro-
grammes wird erneut überprüft, wie viele Stackbereiche aktuell existieren. Nehmen wir nun an, dass 
unser Programm nicht im Originärstack begann (Stack1), sondern seine Daten im Stack 3 hinter-
legte. Nehmen wir weiter an, das von uns gerufene Programm erzeugte seinerseits einen neuen 
Stack, ohne ihn wieder aufzulösen. In diesem Fall hätte unser erster QSTACK-Befehl eine 3 über 
RC geliefert, die in der Variablen AKT_STAK abgelegt wurde. Wir hätten so mit Aufruf des Unter-
programmes den Stack4 gebildet. Wenn alle untergeordneten Programme nun zwei Stacks erzeug-
ten, ohne sie wieder aufzulösen, liefert der zweite QSTACK-Befehl über die Variable RC eine 6. Im 
Anschluss wird eine Schleife RC-AKT_STAK mal (also dreimal) durchlaufen. In der Schleife werden 
die überzähligen Stacks gelöscht, wir landen wieder zurück im Stack3, von dem aus alles begonnen 
hat. Leider kann immer nur im aktuellen Stack gearbeitet werden. 

1.23. Dateiverarbeitung 

In der Dateiverarbeitung unter REXX liegt einer der deutlichsten Vorteile gegenüber CLIST. Eine 
Datei, die unter REXX verarbeitet werden soll, muss nicht explizit geöffnet werden. Ist die Datei zum 
Zeitpunkt des ersten Zugriffes noch geschlossen, wird sie automatisch geöffnet. Der Service des 
dafür zuständigen EXECIO-Befehles geht so weit, dass man prinzipiell sogar bewusst auf das 
Schließen einer Datei verzichten kann. Bei Programmende werden alle geöffneten Dateien ord-
nungsgemäß geschlossen. Aber eine derart schlampige Weise der Programmierung wollen wir gar 
nicht erst aufkommen lassen. 

Der Zugriff auf eine Datei kann über drei verschiedene Attribute erfolgen: 

▪ DISKR 
Lesen aus einer Datei 

▪ DISKW 
Schreiben in eine Datei 

▪ DISKRU 
Kommt von Read and Update. Wird eine Datei mit diesem Attribut geöffnet, wird beim Lesen 
ein Datensatzzeiger mitgeführt, der immer auf den letzten gelesenen Datensatz zeigt. Erfolgt 
im Anschluss an einen Lesebefehl mit dem Attribut DISKRU ein Schreibbefehl (Schreiben 
und Lesen in einer Verarbeitung ist nur möglich, wenn das Lesen über DISKRU durchgeführt 
wird), wird der zu schreibende Datensatz physisch an die Stelle zurückgebracht, von der das 
letzte Lesen stattfand. 

Um eine Datei verarbeiten zu können, muss sie natürlich dem Programm gegenüber bekanntgege-
ben werden. Diese Aufgabe erfüllt für uns der TSO-Befehl ALLOCATE. Er weist einer Datei einen 
symbolischen Namen zu, über den der EXECIO-Befehl auf die Datei zugreift. ALLOCATE ist somit 
das Bindeglied zwischen dem symbolischen Dateinamen in einem Programm und den definitiven 
Datenbeständen, die damit verarbeitet werden sollen. Dies hat den Vorteil, dass ein Programm 
selbst nicht geändert werden muss, wenn es mit unterschiedlichen Datenbeständen zu arbeiten hat. 
Ändern muss sich nur die entsprechende Zuweisung. 

Bevor wir uns mit den Möglichkeiten des EXECIO-Befehles auseinandersetzen, wollen wir uns des-
halb ein wenig mit der Zuweisung und Freigabe symbolischer Dateinamen auseinandersetzen. Dies 
fördert die spätere Erkennung von Problemsituationen. 

Wird eine LOGON-Session aufgebaut, erfolgt dies über eine JCL-Verarbeitung. Im Zuge dieser Job-
Control-Verarbeitung werden alle Dateien, die im Verlauf einer TSO-Sitzung benötigt werden, über 
entsprechende DD-Anweisungen zugewiesen. Jede Zuweisung durch ein DD-Statement führt zu 
einem Eintrag in der adressraum-spezifischen Dateizuweisungstabelle. Diese Tabelle verändert sich 

nun permanent im Laufe einer TSO-Sitzung und kann über den TSO-Befehl LISTA am Terminal 
betrachtet werden. Dieser Befehl kennt eine ganze Reihe möglicher Operanden. Die gängigste Form 
des Aufrufes ist LISTA ST (LIST Allocations with STatus). In dieser Form werden die Namen aller 
seit LOGON zugewiesenen und nicht wieder freigegebenen Dateien mit ihren symbolischen Namen 
und der Dateidisposition ausgegeben. 

Wenn wir uns jetzt so eine Zuweisungstabelle ansehen, wollen wir hier nicht im Detail erörtern, wo 
sie steht (irgendwo im TSO-Adressraum), oder wie sie bis auf das letzte Byte genau aussieht. Wich-
tig ist alleine ein Verständnis für die Arbeitsweise des ALLOCATE- und FREE-Befehles zu bekom-
men. Also ganz oberflächlich: Im Wesentlichen werden der symbolische Name einer Datei und der 
Name eines Datenbestandes, der über diesen symbolischen Namen angesprochen werden soll, 
eingetragen. Sollen unter einem symbolischen Namen mehrere Dateien angesprochen werden (un-
ter dem Symbol SYSEXEC wird eine REXX-Programmbibliothek - gegebenenfalls auch mehrere - 
bekanntgegeben), können entsprechend mehrere Datenbestände unter einem symbolischen Na-
men in die Zuweisungstabelle eingetragen werden. Wir sprechen in diesem Fall von Dateiverknüp-
fung, oder wie es so schön in neudeutsch heißt: Dataset concatination. Angenommen, unsere Da-
teizuweisungstabelle sieht folgendermaßen aus: 

Symbolik Datei 

SYSPROC SYSTEM.MVS.DUMMY 
SYSTEM.TOOLS.CLIST 
SYSTEM.MACRO.CLIST 
SP1.FUNCTION.CLIST 

SYSEXEC SYSTEM.TOOLS.EXEC 
SP1.PRIVATE.EXEC 
SP1.MACRO.EXEC 

ISPPLIB … 

ISPSLIB … 

ISPTLIB … 

ISPFILE … 

ISPTABL … 

ISPPROF … 

 

Wir wollen das Member TEL aus der Datei SP1.EDU.DATA unter dem symbolischen Namen TEL 
unserem Programm zur Verfügung stellen. Dies erfordert eine Zuweisung im Programm, die folgen-
dermaßen aussieht: 

/* REXX  

*/ 

"ALLOC DD(TEL) DA('SP1.EDU.DATA(TEL)') SHR" 

ALLOCATE versucht unter dem symbolischen Namen TEL einen Eintrag in die Dateizuweisungsta-
belle zu machen und eine Verbindung zwischen dem Symbol TEL und der Datei 
SP1.EDU.DATA(TEL) herzustellen. Dieser Eintrag in der Zuweisungstabelle ist nur möglich, wenn 
noch keiner mit dem symbolischen Namen TEL existiert. ALLOC führt in dieser Form nicht zum 
Überschreiben bereits existierender Einträge. Nachdem unter dem Symbol TEL noch kein Eintrag 
existiert, kann ALLOCATE den Eintrag vornehmen (ALLOC liefert einen RC 0). Damit der Eintrag 
nun auch bei zweitem oder drittem Ablauf des Programmes immer wieder erfolgen kann, muss bei 
Programmende dafür gesorgt werden, dass das Symbol TEL wieder freigegeben wird: 

/* REXX  

*/ 

"ALLOC DD(TEL) DA('SP1.EDU.DATA(TEL)') SHR" 

: 

/* Verarbeitung */ 

: 

"FREE DD(TEL)" 

EXIT 

Nun wäre denkbar, dass das Programm aufgrund eines Fehlers ein vorzeitiges Ende findet. ALLOC 
ist ausgeführt, FREE konnte wegen Abbruchs nicht erfolgen. Startet man das Programm erneut, 
wird der ALLOC-Befehl seine Zuweisung nicht durchführen können, weil ein Eintrag unter dem sym-
bolischen Namen TEL noch in der Zuweisungstabelle existiert. ALLOC wird mit RC 8 beendet. Das 
Programm wird beim nächsten Befehl fortgesetzt. Wird immer die gleiche Datei zugewiesen, hat das 
keine problematischen Auswirkungen. Aber wann ist das schon mal so?? 

Ändern wir also unser Programm so, dass der Name der zuzuweisenden Datei vom Anwender ein-
gegeben werden kann, und unter dem gleichen symbolischen Namen bei jeder Verarbeitung eine 
andere Datei, beziehungsweise ein anderes Member der gleichen Datei zugewiesen werden kann: 

/* REXX  

*/ 

SAY "Name der Eingabe (ENTER = Ende)" 

DO FOREVER 

   PARSE UPPER EXTERNAL datei 

   IF SYSDSN(datei) = "OK" THEN LEAVE 

   SAY "Datei" datei "nicht verfügbar, bitte neu" 

END 

"ALLOC DD(TEL) DA("datei") SHR" 

/* Verarbeitung */ 

"FREE DD(TEL)" 

EXIT 

Nun noch einmal zu unserem Problem: Das Programm läuft an. Die Eintragung in die Zuweisungsta-
belle ist erfolgt. Vor dem FREE-Befehl gibt es Programmabbruch. Der Eintrag in der Zuweisungsta-
belle bleibt erhalten und zeigt auf die Datei der ersten Verarbeitung. 

Das Programm soll nochmals gestartet werden, diesmal aber mit einer anderen Eingabedatei. AL-
LOC endet mit RC 8, weil in der Zuweisungstabelle ein Eintrag unter dem Synonym TEL bereits 
existiert, das Programm läuft weiter. Sobald der Zugriff auf die Datei über das Synonym erfolgen 
soll, wird aber nicht die Datei des zweiten Durchlaufes, sondern die des ersten angesprochen. So-
fern das Programm geregelt beendet wird, bleibt dies oft unerkannt und kann zu einem späteren 
Zeitpunkt fatale Folgen haben. Derartige Fehler passieren nach Murphy häufig bei Dateifortschrei-
bungen über längere Zeiträume und dort meist am Anfang eines Verarbeitungszyklus. Diese Situa-
tion kann auf zweierlei Wege umgangen werden. Entweder bauen wir den FREE-Befehl zusätzlich 
vor den ALLOC-Befehl in das Programm (dort würde er nach einem Programmabbruch bei Wieder-
anlauf seine Arbeit verrichten. Ohne vorausgegangenen Abbruch - das Symbol ist in der Zuwei-
sungstabelle nicht eingetragen - liefert der FREE-Befehl hier RC größer 0, der aber nur mitteilt: keine 
Freigabe, da nichts zugewiesen war). Oder wir benutzen den ALLOC-Befehl mit dem Operanden 
REUSE (hat nichts mit Fische fangen zu tun, sondern meint RE-USE (der bestehende Eintrag wird 
überschrieben)). REUSE ist im ALLOC-Befehl aber nur mit einer Zuweisung SHR möglich (Share 
erlaubt Mehrfachzugriff). Deshalb im Anschluss beide Varianten: 

/* REXX *****************************************/ 

SAY "Name der Eingabe (ENTER = Ende)" 

DO FOREVER 

   PARSE UPPER EXTERNAL datei 

   IF SYSDSN(datei) = "OK" THEN LEAVE 

   SAY "Datei" datei "nicht verfügbar, bitte neu" 

END 

"FREE DD(TEL)" 

"ALLOC DD(TEL) DA("datei") SHR" 

/* Verarbeitung */ 

"FREE DD(TEL)" 

EXIT 

 

/* REXX *****************************************/ 

SAY "Name der Eingabe (ENTER = Ende)" 

DO FOREVER 

   PARSE UPPER EXTERNAL datei 

   IF SYSDSN(datei) = "OK" THEN LEAVE 

   SAY "Datei" datei "nicht verfügbar, bitte neu" 

END 

"ALLOC DD(TEL) DA("datei") SHR REUSE" 

/* Verarbeitung */ 

"FREE DD(TEL)" 

EXIT 

Abschließend noch eine Anmerkung zum FREE-Befehl. FREE ist ein sehr zeitintensiver Befehl. 
Wann immer ihr mehrere symbolische Namen freizugeben habt, erledigt das Ganze möglichst immer 
in einem einzigen Befehl. Fünf FREE-Befehle, die nacheinander einen symbolischen Namen freige-
ben, dauern beinahe fünfmal so lange wie einer, der alle fünf Namen auf einmal freigibt: 

"FREE DD(name1 name2 name3 name4 name5)" 

Betrachten wir jetzt den verschiedenen Möglichkeiten der Schreib-/Leseoperationen unter REXX. 

1.23.1. Lesen aus einer Datei 

Das Lesen aus einer Datei hat als Outputdefinition im Regelfall den Stackbereich. Der entspre-
chende EXECIO-Befehl in seiner gängigsten Form leistet hier phänomenales. Beim ersten Zugriff 
wird geprüft, ob die Datei geöffnet ist. Wenn nicht, erfolgt der OPEN auf den Datenbestand, ohne 
dass wir uns selbst darum kümmern müssen. Anschließend werden in Schleife alle Datensätze der 
Datei gelesen und in den Stack kopiert, bis EOF in der Eingabedatei erkannt wird. Die Datei wird 
geschlossen, die Leseverarbeitung ist abgeschlossen. Sehen wir uns hierzu ein kleines Beispiel an: 

/* REXX * ***************************************/ 

"ALLOC DD(EIN) DA('SP1.EDU.DATA(TEL)') SHR REUSE" 

"EXECIO * DISKR EIN (FINIS" 

"FREE DD(EIN)" 

Die Datensätze des Members TEL der SP1.EDU.DATA liegen im Stack. Das Lesen ist abgeschlos-
sen, die Eingabedatei wieder freigegeben. Um die Datensätze auszugeben, muss noch der Stack-
bereich abgearbeitet werden. In Schleife (sooft, als Sätze im Stack liegen - diese Zahl QUEUED()) 
wird je ein Datensatz aus dem Stack entnommen und am Terminal ausgegeben: 

DO QUEUED() 

   PARSE PULL satz 

   SAY satz 

END 

Wenn möglich, sollte die Ausgabe der EXECIO-Leseoperation immer der Stack sein. Stackverarbei-
tung ist grundsätzlich schneller als das Hinterlegen der gelesenen Sätze in Speichervariablen. 
Manchmal ist es von der Programmlogik her aber nicht sinnvoll. Sei es, weil im Anschluss ein wahl-
freier Zugriff auf die einzelnen Datensätze erfolgen soll (der Stack kann nur sequentiell verarbeitet 
werden), weil im Stack bereits Daten liegen, die zu einem späteren Zeitpunkt verarbeitet werden 
sollen, oder weil wir mehrere Datenbestände parallel zueinander verarbeiten möchten (Dateiver-
gleich oder Mischverfahren). 

Gehen wir vom oben angeführten Beispiel aus und hinterlegen die gelesenen Datensätze in einem 
Variablenstamm. Der EXECIO-Befehl ändert sich nur unwesentlich: 

/* REXX *****************************************/ 

"ALLOC DD(EIN) DA('SP1.EDU.DATA(TEL)') SHR REUSE" 

"EXECIO * DISKR EIN (STEM satz. FINIS" 

"FREE DD(EIN)" 

Durch die Angabe STEM satz. im EXECIO-Befehl werden die gelesenen Sätze nicht im Stack ab-
gelegt. Stattdessen wird ein Stem aufgebaut. Der erste gelesene Satz wird in SATZ.1 hinterlegt. Der 
zweite Satz in SATZ.2 und so weiter. Soll dieser Variablenstamm nun ausgegeben werden, ist wich-
tig zu wissen, wie viele Datensätze gelesen wurden, das heißt, wie viele Variablen SATZ.? erzeugt 
wurden. Die QUEUED()-Funktion bringt uns an dieser Stelle nicht weiter, weil sie ja stackbezogene 
Information liefert. EXECIO ist so intelligent, nach Abschluss der Leseoperation in der .0-Variablen 
des Stammes zu hinterlegen, wie viele Datensätze gelesen, beziehungsweise wie viele Variablen 
innerhalb des Stammes gebildet wurden. Die Schleife der Bildschirmausgabe sieht dementspre-
chend auch etwas anders aus. Eine Möglichkeit wäre DO satz.0 (sooft, wie Sätze gelesen wur-

den). Wir müssten dann aber einen Zähler initialisieren und in der Schleife erhöhen. Er erlaubt uns 
zu Beginn der Verarbeitung an die erste Variable des Stammes zu greifen. In der Schleife wird bei 
jedem Durchlauf durch Erhöhung des Zählers auf die jeweils nächste Variable zugegriffen. Hierfür 
haben wir eine Schleifenvariante, die sich selbst um die Erhöhung des Zählers kümmert: 

DO idx=1 TO satz.0 

   SAY satz.idx 

END 

EXIT 

In beiden Fällen gehen wir davon aus, dass die Datei in einem Stück im TSO-Adressraum abgebildet 
werden kann. Sei es nun im Stack oder in Form von Compound-Variablen. Bei sehr großen Dateien 
ist dies unter Umständen nicht möglich. Ist die Größe der LOGON-Session nicht ausreichend, um 
eine Datei gesamt abzubilden, muss entweder die Adressraumgröße beim LOGON anders gewählt 
werden (LOGON userid SIZE(8192) liefert uns einen TSO-Adressraum in der Größe von 8MB), 

oder die Datei muss abschnittweise gelesen werden. Wenngleich man bei Verarbeitung solch großer 
Dateien unter REXX mal darüber nachdenken sollte, ob es nicht eine bessere Möglichkeit gäbe, 
sehen wir uns im nächsten Beispiel an, wie dieses päckchenweise Lesen einer Datei aussehen 
kann: 

/* REXX *****************************************/ 

"ALLOC DD(EIN) DA('SP1.EDU.DATA(TEL)') SHR REUSE" 

DO FOREVER 

   "EXECIO 200 DISKR EIN" 

   datei_ende = (RC = 2) 

   DO QUEUED() 

      PARSE PULL satz 

      SAY satz 

   END 

   IF datei_ende THEN LEAVE 

END 

"EXECIO 0 DISKR EIN (FINIS" 

"FREE DD(EIN)" 

EXIT 

Nachdem der Lesebefehl wiederholbar sein muss, liegt er in einer Schleife. Der Einfachheit halber 
kodieren wir die Schleife DO FOREVER. EXECIO legt 200 Datensätze im Stackbereich ab. Die 
Klausel (FINIS darf auf keinen Fall kodiert werden (EXECIO würde die Datei sonst öffnen, 200 

Datensätze lesen und in den Stack stellen und die Datei wieder schließen. Beim nächsten Schlei-
fendurchlauf würde die Datei wieder geöffnet werden, 200 Sätze würden gelesen (wieder die ersten 
200)...).So bleibt die Datei nach dem Lesen der ersten 200 Sätze geöffnet. Wird EXECIO im zweiten 
Durchlauf verarbeitet, werden die nächsten 200 Datensätze verarbeitet. Das Lesen in Schleife wird 
irgendwann zum Erkennen des EOF-Ereignisses führen. EXECIO meldet dieses Ereignis über RC 
2. Wir dürfen im direkten Anschluss nach EXECIO aber die Verarbeitung auch bei aufgetretenen RC 
2 nicht beenden. Es ist davon auszugehen, dass vor dem Erkennen des EOF-Ereignisses noch 
einige Datensätze gelesen werden konnten. Diese Sätze stehen noch im Stack und müssen verar-
beitet werden, bevor die Leseschleife verlassen werden darf. Aus diesem Grund wird, wenn EXECIO 
Dateiende erkennt, zunächst nur ein Schalter für Verarbeitungsende (S_ENDE=1) auf aktiv gesetzt. 
Nachdem die letzten gelieferten Datensätze verarbeitet wurden, wird der Endeschalter abgefragt. 
Ist er aktiv(1), wird die Schleife verlassen. Die Datei ist zu diesem Zeitpunkt immer noch geöffnet. 
Auch wenn es REXX-seitig nicht unbedingt nötig wäre, das Schließen einer Datei bei Verarbeitungs-
ende gehört ganz einfach zu einer ordentlichen Programmierung. Wir erledigen dies im EXECIO-
Befehl außerhalb der Schleife (zu verarbeitende Datensätze 0, durch FINIS wird lediglich die Datei 
geschlossen). 

Sonderfall: Nehmen wir an, in der Datei liegen genau 200 Datensätze. Nachdem EXECIO die ersten 
200 Sätze gelesen hat, ist noch nicht erkennbar, dass mit dem nächsten Leseversuch nur noch EOF 
erkannt werden kann. Der gemeldete Return Code des EXECIO ist dementsprechend 0. Die gelie-
ferten 200 Datensätze werden über die Schleife DO QUEUED() ausgegeben. Wird EXECIO nun 
zum zweiten Mal ausgeführt, endet der Befehl mit RC 2, ohne noch weitere Datensätze im Stack 
abgestellt zu haben. Für unsere Verarbeitung spielt dies keine Rolle. Ist der Stack leer, führt DO 
QUEUED() zu keiner Schleifenverarbeitung mehr (QUEUED() liefert den Wert 0, DO 0 würde be-
deuten: durchlaufe die Schleife 0 mal). Wird die innere Schleife in diesem Fall übergangen, treffen 
wir entsprechend wieder auf die Abfrage des Endeschalters und würden die äußere Verarbeitungs-
schleife aufgrund der erfüllten Bedingung an dieser Stelle verlassen. 

1.23.2. Schreiben in eine Datei 

So, wie wir bei der Leseoperation des EXECIO den TSO-Adressraum als Ausgabe haben (egal, ob 
in Form des Stacks oder als Compound-Variable), nutzen wir ihn bei Schreiboperationen als Ein-
gabe. Auch hier können wir zwischen den beiden Varianten Stack und Variablenstamm wählen. 
Bevor wir die Schreiboperation durchführen können, ist es erforderlich, den zu schreibenden Daten-
bestand im Stack oder als Variablenstamm aufzubereiten. 

Sehen wir uns die beiden Varianten der Reihe nach an. Beginnen wir mit dem Stack: 

Sind die zu schreibenden Sätze im Stack, dient dieser dem EXECIO als "Inputdatei". Beim Lesen 
der Eingabe aus einer normalen Datei würde nach dem letzten Datensatz EOF erkannt werden, das 
Lesen würde beendet werden. Im Stack haben wir nun kein EOF im üblichen Sinn. EXECIO inter-
pretiert EOF im Stack, wenn er auf einen NULLSTRING trifft. Nullstring im Stack wird erzeugt, wenn 
ein QUEUE-Befehl kodiert wird, hinter dem weder ein Literal noch eine Variable folgt. Wird versäumt, 
diesen Nullstring im Stack zu hinterlegen, ergibt sich für den Anwender eine sehr eigenartige Situa-
tion: 

Wie PARSE PULL kann auch EXECIO unmittelbar von der Terminaltastatur lesen. Sind alle Sätze 
aus dem Stack entnommen worden, ohne dass EXECIO Nullstring erkannt hat, wendet er sich zum 
weiteren Lesen an die Tastatur. Für den Anwender, der davon nichts weiß, äußert sich das Ganze 
wie Programmstillstand. Nachdem die meisten EDV-ler sehr ungeduldige Menschen sind, wird der 
Anwender nach einiger Zeit entnervt auf seine Tastatur einhämmern. Was immer er dabei tippt, nach 
<ENTER> wird es von EXECIO als Datensatz übernommen und in die Ausgabedatei geschrieben. 
Gelingt es dem Anwender zufällig, nur die Taste <ENTER> zu treffen, ohne vorher Daten zu tippen, 
empfängt der EXECIO-Befehl auf diese Weise seinen Nullstring, den er als EOF interpretiert, und 
die Schreibverarbeitung kann beendet werden. Ein Programm, das auf diese Weise arbeitet, ver-
breitet bei den Anwendern sehr viel Freude. 

Das folgende Beispiel zeigt eine Verarbeitung, bei der Daten aus einem Member einer PO-Datei 
gelesen werden. Der Name des Members ist variabel und wird am Terminal abgefragt. Diese Daten 
werden mit dem Hinweis auf Datum und Uhrzeit an das Ende einer sequentiellen Datei angehängt. 

/* REXX *****************************************/ 

SAY "Eingabe-Member (ENTER = Ende)" 

DO FOREVER 

   PULL mbr 

   IF LENGTH(mbr) = 0 THEN EXIT 

   IF SYSDSN("EDU.DATA("mbr")") = "OK" THEN LEAVE 

   SAY "Member" mbr "nicht verfügbar, bitte neu" 

END 

"ALLOC DD(EIN) DA(EDU.DATA("mbr")) SHR REUSE" 

"ALLOC DD(AUS) DA(SEQSTAT.DATA) MOD" 

"EXECIO * DISKR EIN (FINIS" 

IF QUEUED() > 0 THEN DO 

   QUEUE 

   PUSH "Daten " mbr "vom" date() "um" time() 

   "EXECIO * DISKW AUS (FINIS" 

END 

"FREE DD(EIN AUS)" 

EXIT 

Die Prüfung IF QUEUED() > 0 lässt uns erkennen, ob das Lesen des ausgewählten Members Daten 
geliefert hat oder nicht. In diesem Fall wäre das Member inhaltlich leer gewesen (soll es auch schon 
mal gegeben haben). Nur wenn die Leseoperation Daten im Stack abgeliefert hat, wird der Nullstring 
angehängt, die Hinweiszeile vor die Datensätze eingeschoben und der Inhalt des Stackbereiches in 

seiner Gesamtheit an das Ende der sequentiellen Datei userid.SEQSTAT.DATA angefügt. Auf diese 
Weise haben wir für die Fortschreibung der Daten gesorgt. 

ACHTUNG: Fortschreibung mit Disposition MOD ist nur für sequentielle Dateien zulässig. Für Mem-
ber aus PO-Dateien ist dies nicht erlaubt. Aber wenn wir schon dabei sind, schauen wir uns gleich 
an, wie das Problem zu lösen wäre, wenn die Fortschreibung der Daten in einem Member STATIST 
der gleichen PO-Datei durchzuführen wäre: 

/* REXX *****************************************/ 

SAY "Eingabe-Member (ENTER = Ende)" 

DO FOREVER 

   PULL mbr 

   IF LENGTH(mbr) = 0 THEN EXIT 

   IF SYSDSN("EDU.DATA("mbr")") = "OK" THEN LEAVE 

   SAY "Member" mbr "nicht verfügbar, bitte neu" 

END 

"ALLOC DD(EIN) DA(EDU.DATA("mbr")) SHR REUSE" 

"ALLOC DD(STAT) DA(EDU.DATA(STATIST)) OLD" 

"EXECIO * DISKR STAT (FINIS" 

QUEUE "Daten aus" mbr "vom" date() "um" time() 

stat_rec = QUEUED() 

"EXECIO * DISKR EIN (FINIS" 

IF QUEUED() > stat_rec THEN DO 

   QUEUE 

   "EXECIO * DISKW STAT (FINIS" 

END 

"FREE DD(EIN STAT)" 

EXIT 

Hier sind wir gezwungen, die Statistikdaten vorzulesen und im Stack abzulegen. Das Statistik-Mem-
ber wird anschließend wieder geschlossen. Jetzt wird prophylaktisch die Hinweiszeile an die Altda-
ten angehängt, und wir merken uns, wie viele Datensätze zum jetzigen Zeitpunkt im Stack liegen. 
Nun können wir die Daten des ausgewählten Members hinzu lesen. Sind die Anzahl Datensätze im 
Stack immer noch gleich viele, wie in STAT_REC hinterlegt, war das betreffende Member leer. An-
derenfalls wird Nullstring in den Stack geschrieben und der gesamte Stackinhalt in das Statistik-
member überführt. Die Altdaten des Statistikmembers werden dabei überschrieben. 

Für die zweite Variante der Schreiboperation konstruieren wir uns folgendes Beispiel 

Ein Bediener soll in einer Erfassungsschleife die Möglichkeit zu beliebig vielen Eingaben bekommen. 
Jede dieser Eingaben soll eine Compound-Variable füllen. Wird bei der Eingabeaufforderung nur 
<Enter> gedrückt, ist Erfassungsende, und der in der Schleife gebildete Variablenstamm wird in eine 

Datei geschrieben. Wir setzen im nachfolgenden Beispiel eine Schleife ein, die wir in dieser Form 
noch nicht angewandt haben. Sie kommt nicht sehr häufig vor, liefert uns aber in dieser speziellen 
Situation einen Vorteil. Und überhaupt: Beispiele kann man nie genug sehen. Entscheidet euch 
selbst für die eine oder andere Variante. 

/* REXX *****************************************/ 

SAY "Dateneingabe (ENTER = Ende)" 

DO cnt=1 

   SAY cnt". Eingabe" 

   PARSE EXTERNAL eingabe.cnt 

   IF LENGTH(eingabe.cnt) = 0 THEN LEAVE 

END 

DROP eingabe.cnt 

cnt = cnt - 1 

"ALLOC DD(AUSG) DA(TEST.DATA) OLD" 

"EXECIO * DISKW AUSG (STEM eingabe. FINIS" 

"FREE DD(AUSG)" 

EXIT 

Über den Schleifenansatz DO cnt=1 würden wir eine Endlosschleife initiieren, bei der CNT mit 1 
vorbelegt wird und sich mit jedem Durchlauf um 1 erhöht. Diese Form wird in der offiziellen Literatur 
nicht propagiert. Der Zähler wird genutzt, um den Aufbau eines Stems zu ermöglichen. Das Schlei-
fenende wird über die Längenprüfung der Eingabevariablen gesteuert. Ergibt die Prüfung Länge 0, 
wird die Schleife verlassen. 

Hier müssen wir uns nun aber detailliert ansehen, was passiert. LEAVE führt nicht zum Absprung 
aus der Schleife. Wurden vom Bediener drei gültige Eingaben gemacht und beim vierten Durchlauf 
nur E gedrückt, sind die Variablen EINGABE.1 bis .3 entsprechend mit der Eingabe des Bedieners 

gefüllt, .4 ist ein Nullstring, .5 wurde nicht genutzt und zeigt auf sich selbst. Warum diese Situation 
nicht gut für unseren weiteren Ablauf ist, wird verständlich, wenn wir uns das Verfahren des EXECIO 
(Lesen aus Variablenstamm, Schreiben in Datei) genauer ansehen: 

Füllt EXECIO einen Variablenstamm durch die Leseoperation, wird in der jeweiligen .0-Variablen 
hinterlegt, wie viele Datensätze geliefert wurden. Beim Schreiben wäre nun naheliegend, dass wir 
im Programm vor der Schreiboperation in der zuständigen .0-Variablen hinterlegen müssen, wie 
viele Variablen der Stamm umfasst, der geschrieben werden soll. Dem ist nicht so. EXECIO verfährt 
hier grundlegend anders. Es wird davon ausgegangen, dass der erste zu schreibende Datensatz in 
der Variablen .1 liegt, der zweite in .2 und so weiter. EXECIO tastet sich durch das Erhöhen des 
Stammzeigers so von einer Variablen zur anderen. Dies wird irgendwann dazu führen, dass eine 
Variable .irgendwas im Programmverlauf nicht gebildet wurde und deshalb inhaltlich auf ihren Ei-
gennamen zeigt. Diese Situation wird von EXECIO als Pseudo-EOF innerhalb des Variablenstam-
mes gewertet.  

Auf unser Beispiel bezogen würde dies bedeuten, die Variable EINGABE.4 (Nullstring) wird als Leer-
zeile in der Datei abgelegt. Wir hätten einen (ungewollten) Datensatz zu viel geschrieben. Dieses 
Problem lösen wir durch das Rücksetzen der letzten erzeugten Variablen des Stammes (DROP 
EINGABE.CNT). Die Verminderung der Variablen CNT um eins wäre für die Richtigkeit unseres 
Programmes nicht nötig, ist aber dennoch sinnvoll. Auf diese Weise könnte zu einem späteren Zeit-
punkt im Programm über CNT festgestellt werden, wie viele gültige Eingaben in der Erfassungs-
schleife gemacht wurden (Man soll Arbeit verhindern, wo immer es möglich ist. Ist sie aber nicht 
vermeidbar, ist ordentliche Arbeit im Allgemeinen auch kein größerer Aufwand als schlampige). 

Alternativ dazu böte sich folgende Lösung an: 

/* REXX *****************************************/ 

SAY "Dateneingabe (ENTER = Ende)" 

cnt = 0 

DO FOREVER 

   SAY cnt+1". Eingabe" 

   PARSE EXTERNAL eingabe 

   IF LENGTH(eingabe) = 0 THEN LEAVE 

   cnt = cnt + 1 

   eingabe.cnt = eingabe 

END 

"ALLOC DD(AUSG) DA(TEST.DATA) OLD" 

"EXECIO * DISKW AUSG (STEM eingabe. FINIS" 

"FREE DD(AUSG)" 

EXIT 

Verbinden wir Lesen und Schreiben in einem Programm, können wir uns mit wenig Aufwand ein 
kleines Kopier-Utility basteln. Wenn wir hier auf Plausibilitätsprüfungen verzichten (was man in der 
Praxis natürlich nicht täten), ist immer wieder verblüffend, mit wie wenig Aufwand manche Dinge 
erreichbar sind. Das Programm spricht für sich selbst: 

/* REXX *****************************************/ 

ARG in out 

"ALLOC DD(EIN) DA("in") SHR REUSE" 

"ALLOC DD(AUS) DA("out") OLD" 

"EXECIO * DISKR EIN (FINIS" 

"EXECIO" QUEUED() "DISKW AUS (FINIS" 

"FREE DD(EIN AUS)" 

EXIT 

1.23.3. Update (Lesen und Schreiben in einem Verarbeitungszyklus) 

Die logische Konsequenz aus Lesen und Schreiben ist der UPDATE (die Veränderung eines Daten-
satzes in der Datei, ohne die Gesamtdatei Rückschreiben zu müssen). Damit dies möglich ist, wird 
vorausgesetzt: 

▪ Dem Schreiben in die Datei muss ein Lesebefehl vorausgehen. 
▪ Die Datei darf zunächst nicht geschlossen werden. 
▪ Der Lesebefehl muss mit einem Attribut DISKRU (Read and Update) kodiert werden. 
▪ Das Lesen muss satzweise geschehen. Auf diese Weise wird beim Lesen ein Satzzeiger in 

der Datei mitgeführt, der immer auf den zuletzt gelesenen Datensatz zeigt. 
▪ Erfolgt im Anschluss an einen Lesebefehl ein Schreibbefehl, ist aufgrund des Satzzeigers 

klar, welcher Datensatz in der Datei überschrieben werden soll (es ist der Satz, auf den der 
Satzzeiger zeigt - der zuletzt gelesene). 

Auch hierzu zwei kleine Beispiele. Im ersten Beispiel gehen wir davon aus, dass wir es bei der 
Änderung eines Datensatzes mit einem eindeutigen Suchkriterium zu tun haben (Personalnummer). 
Weiter unterstellen wir, dass die Personalnummer im Datensatz auf Stelle 1 bis 4 und die Telefon-
nummer des Mitarbeiters auf Stelle 101 bis 120 steht. Wir suchen nach einer bestimmten Personal-
nummer und führen die Änderung der Telefonnummer durch. 

Das satzweise Lesen integrieren wir in eine FOREVER-Schleife. Wird der gesuchte Datensatz ge-
funden und geändert, können wir die Schleife verlassen und geben eine entsprechende positive 
Meldung aus. Wird der Suchbegriff nicht gefunden, muss die Datei komplett gelesen werden. Schlei-
fenende ist dann erreicht, wenn das Lesen eines Datensatzes zum Erkennen des EOF geführt hat. 
In diesem Fall wird die Schleife ebenfalls verlassen und eine negative Meldung ausgegeben. 

Um das Ganze zu vereinfachen, belegen wir uns eine Nachricht mit negativem Text vor (auch wenn 
man grundsätzlich immer positiv denken sollte). Erkennen wir im Programm, dass der gesuchte Satz 
gefunden wurde, führen wir die Änderung durch und ändern die Meldung. 

Wir wollen hier auf die letzten Feinheiten der Plausibilitätsprüfungen verzichten, da wir ja in erster 
Linie den Update der Datei veranschaulichen wollen. Unsere Prüfungen, die im Programm durchge-
führt werden, verhindern nur das schlimmste (Personalnummer muss numerisch sein, Telefonnum-
mer wird automatisch auf eine Länge von 20 Byte gestreckt und mit BLANKs gefüllt, oder im Be-
darfsfall auf 20 Byte verkürzt. Ist die Personaldatei nicht zu haben, wird die Verarbeitung nach Aus-
gabe einer Hinweismeldung vorzeitig beendet). 

/* REXX * UPDDEMO1 ******************************/ 

DO UNTIL DATATYPE(pnum) = "NUM" 

   SAY "Personalnummer eingeben (ENTER = Ende)" 

   PULL pnum 

   IF LENGTH(pnum) = 0 THEN EXIT 

END 

"ALLOC DD(PERS) DA('GEHEIM.PERSONAL.DATA') OLD" 

IF rc > 0 THEN DO 

   SAY "Zuweisung fehlerhaft, RC:" rc 

   SAY "Bitte prüfen!!" 

   EXIT 

END 

msg="Personalnummer" pnum "nicht gefunden" 

DO FOREVER 

   "EXECIO 1 DISKRU PERS" 

   IF rc = 2 THEN LEAVE 

   PARSE PULL satz 

   IF LEFT(satz,4) = pnum THEN DO 

      msg="Personalnummer" pnum "gefunden." 

      SAY "Alte Telefonnummer:" SUBSTR(satz,101,20) 

      SAY "Neue Telefonnummer eingeben" 

      PARSE EXTERNAL tnum 

      IF LENGTH(tnum) > 0 THEN DO 

         msg=msg "Änderung durchgeführt" 

         satz=OVERLAY(LEFT(tnum,20)satz,101) 

         QUEUE satz 

         "EXECIO 1 DISKW PERS" 

         LEAVE 

      END 

      ELSE msg=msg "Keine Änderung" 

   END 

END 

"EXECIO 0 DISKW PERS (FINIS" 

"FREE DD(PERS)" 

SAY msg 

EXIT 

Im zweiten Beispiel wollen wir eine Artikeldatei verarbeiten. Unterstellen wir, dass die ersten sechs 
Byte des Satzes die Artikelnummer enthalten, davon die ersten beiden Byte eine Artikel-Stamm-
nummer. Wir wollen den Artikelstamm 24 auf 34 ändern. 

/* REXX * UPDDEMO2 ******************************/ 

"ALLOC DD(ARTIKEL) DA("ATRIKEL.STAMM.DATA") OLD" 

IF rc > 0 THEN DO 

   SAY "Zuweisung fehlerhaft, RC:" rc 

   SAY "Bitte prüfen!!" 

   EXIT 

END 

c_recupd = 0 

"EXECIO 1 DISKRU ARTIKEL" 

DO WHILE RC = 0 

   PARSE PULL art_satz 

   IF LEFT(art_satz,2) = "24" THEN DO 

      art_satz = "3"SUBSTR(art_satz,2) 

      QUEUE art_satz 

      "EXECIO 1 DISKW ARTIKEL" 

      c_recupd = c_recupd + 1 

   END 

   "EXECIO 1 DISKRU ARTIKEL" 

END 

"EXECIO 0 DISKRU ARTIKEL (FINIS" 

"FREE DD(ARTIKEL)" 

SAY "Anzahl geänderter Artikelsätze:" c_recupd 

EXIT 

Hier haben wir eine andere Programmiertechnik gewählt. Zunächst wird ein Datensatz vorgelesen. 
Würde das erste Lesen in der Datei zu einem Return Code größer 0 führen (es kann dann nur RC 2 
- Dateiende - sein), wird die Schleife übersprungen, die Datei wird geschlossen und freigegeben. 

Kann beim Vorlesen ein Satz aus der Datei empfangen werden, wird die Schleife betreten. Es erfolgt 
eine Prüfung, die uns erkennen lässt, ob es sich bei dem gelesenen Satz um einen handelt, der 
geändert werden soll. Wenn ja, erfolgt die Änderung und ein Zähler für geänderte Artikelsätze wird 
erhöht. 

Als letzter Befehl in der Schleife erfolgt das Nachlesen. Die Ablaufsteuerung verzweigt im Anschluss 
zurück an den Schleifenkopf und prüft die Bedingung erneut. Ist sie erfüllt, hat uns das Nachlesen 
erneut einen Datensatz gebracht, die Schleife wird erneut durchlaufen. Ist die Bedingung unwahr 
(bei EOF im Nachlesen), wird die Schleife verlassen. 

1.23.4. ALLOC: TSO oder UNIX? 

Heute wird in vielen Fällen die ALLOC-Funktion aus BPXWDYN (ein Befehl aus dem USS) dem 
TSO-Alloc vorgezogen. Nicht nur, dass der BPXWDYN-Alloc viel mehr kann, er ist auch viel schöner 
zu kodieren. So ist es beispielsweise mit der BPXWDYN-Variante auch möglich, GDGs zu allokieren 
und bei der Nutzung von temporären Dateien kann sowohl auf die Angabe eines DD-Namens als 
auch auf den Dateinamen verzichtet werden. Beides kann vom Alloc zurückgemeldet werden. 

Die detaillierte Beschreibung ist im Band-2, Syntax-Referenz zu finden. 

1.23.5. Einlesen von Daten in mehrdimensionale Tabelle 

Vorab: Man sollte immer die Konsequenzen seines Handelns im Blick haben. Die Frage ist nicht, 
was alles geht, sondern immer, was am sinnvollsten ist. Dennoch soll das nachfolgende Beispiel 
zeigen, was in Punkto Tabellenlogik in REXX möglich ist (und es ginge noch viel weiter). 

Das Beispiel soll für sich selbst sprechen: 

/* REXX  

*/ 

"ALLOC DD(IN) DA(EDU.DATA(ADR4)) SHR REUSE"  

"EXECIO * DISKR IN (STEM satz. FINIS" 

cnt.   = 0 

allPLZ = 

DO §=1 to satz.0 

   plz = substr(satz.§,30,5) 

   IF wordpos(plz,allPLZ) = 0 THEN allPLZ = allPLZ plz 

   cnt.plz.0 = cnt.plz.0 +1 

   call value "ADR."plz"."cnt.plz.0,satz.§ 

END 

allPLZ = SortWord(allPLZ) /* User Defined Function */ 

DO §=1 to words(allPLZ) 

   plz = word(allPLZ,§) 

   say;say "Ergebnis zu PLZ" plz "("cnt.plz.0")";say 

   do §§=1 to cnt.plz.0 

      say right(§§,4)">" left(adr.plz.§§,72) 

   end 

end 

exit 

 

SortWord: procedure 

   Arg data 

   Data.0 = words(data) 

   Do §=1 to Data.0 

      Data.§ = word(data,§) 

   end 

   call bpxwunix "/bin/sort",data.,data.,stderr. 

   Sorted = 

   Do §=1 to Data.0 

      Sorted = Sorted data.§ 

   end 

return strip(sorted) 

1.24. JOB-CONTROL-Generierung 

Sollen zeitintensive Verarbeitungen in einem Batch-Adressraum ausgeführt werden, bietet sich 
REXX durch seinen Komfort oft für vorbereitende Arbeiten an. REXX kann im Dialog mit dem Be-
diener die variablen Parameter der zu übergebenden Job-Control-Statements erfragen, die Job-
Control-Statements entsprechend ergänzen und an den Internal Reader des Spools übergeben. Der 
Befehl, der die Übergabe durchführt, ist zwar ein TSO-Kommando, in einer bestimmten Schreib-
weise kann es aber nur aus einem REXX-Programm oder einer CLIST benutzt werden. 

Sehen wir uns zunächst die beiden Formen des SUBMIT-Kommandos an: 

SUBMIT datei 

SUBMIT datei(member) 

Diese beiden Formen können auch direkt unter TSO angewandt werden. Die Konventionen in Bezug 
auf Dateinamen unter TSO müssen berücksichtigt werden. 

SUBMIT * END(..) 

Auf diesem Weg kann ein REXX-Programm Job-Control-Statements direkt aus dem Speicher an 
den Internal Reader des Spools übergeben. Diese Variante entspricht der Job-Control-Schreibweise 

//ddname DD * 

und meint: dazugehörendes folgt im Anschluss. Wörtlich genommen trifft dies aber nur für CLIST 
zu. Hier werden die Job-Control-Statements im unmittelbaren Anschluss kodiert. Unter REXX liest 
der SUBMIT-Befehl den Stackbereich aus. Ende der Job-Control-Statements wird erkannt, wenn ein 
Satz gelesen wird, der auf Stelle 1 und 2 die Zeichenkombination führt, wie sie im Parameter END(..) 
angegeben ist. 

Wir sehen uns drei Beispiele an, wie eine Job-Control-Verarbeitung aus einem REXX-Programm 
heraus aussehen kann. Stehen in der Praxis alle üblichen Komponenten des TSO zur Verfügung, 
wird man an dieser Stelle unter Umständen den File-Tailoring-Service des Dialog Managers nutzen, 
sofern man damit umgehen kann. 

Alle drei Beispiele zeigen letztlich die Ausführung des Dienstprogrammes IEBGENER, über das eine 
Druckausgabe durchgeführt werden soll. Unsere kleine REXX-Routine liefert die variablen Parame-
ter und stellt dabei sicher: 

Die zu druckende Datei ist verfügbar, überschreitet die maximale Satzlänge von 132 Byte nicht und 
hat Dateiformat PS oder PO. Im Falle einer gegliederten Datei ist ein Member angegeben. Output-
klasse kann nur A, R oder X sein. 

/* REXX * JCLPRT1 *******************************/ 

DO FOREVER 

   SAY "Druckdatei vollqualifiziert (ohne Hochkomma)" 

   PARSE UPPER EXTERNAL drudat 

   IF LENGTH(drudat) = 0 THEN EXIT 

   IF SYSDSN("'"drudat"'") = "OK" THEN DO 

      DMY = LISTDSI("'"drudat"'") 

      SELECT 

         WHEN POS(SYSDSORG,"PS PO") = 0 THEN 

            msg = "Falsche Dateiorganisation" 

         WHEN SYSDSORG = "PO" & , 

            RIGHT(drudat,1) >< ")" THEN 

            msg = "PO-Datei ohne Memberangabe" 

         WHEN SYSLRECL > 132 THEN 

            msg = "Satzlänge zu groß" 

         OTHERWISE LEAVE 

      END 

   END 

   ELSE msg = drudat "nicht verfügbar" 

   SAY msg 

END 

SAY "Outputklasse (A, R oder X(=Standard))" 

PARSE UPPER EXTERNAL okl 

IF wordpos(OKL,"A R X") = 0 THEN okl = "X" 

QUEUE "//DRUCK    EXEC PGM=IEBGENER" 

QUEUE "//SYSUT1   DD DSN="drudat",DISP=SHR" 

QUEUE "//SYSUT2   DD SYSOUT="okl 

QUEUE "//SYSIN    DD DUMMY" 

QUEUE "//SYSPRINT DD DUMMY" 

QUEUE "##" 

jc_str = "ABCDEFGHIJKLMNOPQRSTUVWXYZ0123456789" 

jcb = SUBSTR(jc_str,random(1,length(jc_str)),1) 

"SUB * END(##) JOBCHAR("jcb")" 

EXIT 

Das erste Beispiel zeigt ein Verfahren, bei dem die Job-Control-Statements durch QUEUE-Befehle 
in den Stackbereich eingearbeitet werden. Dabei werden die starren Teile der Job-Control als Lite-
rale, die flexiblen Teile als REXX-Variable geliefert. Die Variablen werden beim Einstellen in den 
Stack durch ihre Werte ersetzt. Ein Nachteil wäre, wenn man so will, dass die Job-Control-State-
ments selbst fix im Programm hinterlegt sind und Änderungen an der JCL immer Programmände-
rungen in REXX bedingen. 

Hier versuchen wir File-Tailoring nachzuvollziehen. Das Job-Control-Gerippe wird aus einer Datei 
gelesen und in einem Variablenstamm hinterlegt. Danach wird der erzeugte Variablenstamm in den 
Stack eingearbeitet, wobei die Variablen substituiert werden. Dies bedingt die Anwendung der In-
struktion INTERPRET und erscheint auf den ersten Eindruck vielleicht komplizierter, als das voraus-
gegangene Beispiel. Vorteil ist aber, dass das REXX-Programm von JCL-Statements freigehalten 
wird und die Änderungen der JCL einfacher werden: 

/* REXX * JCLPRT2 *******************************/ 

DO FOREVER 

   SAY "Druckdatei vollqualifiziert (ohne Hochkomma)" 

   PARSE UPPER EXTERNAL drudat 

   IF LENGTH(drudat) = 0 THEN EXIT 

   IF SYSDSN("'"drudat"'") = "OK" THEN DO 

      DMY = LISTDSI("'"drudat"'") 

      SELECT 

         WHEN POS(SYSDSORG,"PS PO") = 0 THEN 

            msg = "Falsche Dateiorganisation" 

         WHEN SYSDSORG = "PO" & , 

            RIGHT(drudat,1) >< ")" THEN 

            msg = "PO-Datei ohne Memberangabe" 

         WHEN SYSLRECL > 132 THEN 

            msg = "Satzlänge zu groß" 

         OTHERWISE LEAVE 

      END 

   END 

   ELSE msg = drudat "nicht verfügbar" 

   SAY msg 

END 

SAY "Outputklasse (A, R oder X(=Standard))" 

PARSE UPPER EXTERNAL okl 

IF OKL >< "A" & okl >< "R" THEN okl = "X" 

"ALLOC DD(JCL) DA(SKELS.CNTL(GENER1)) SHR REUSE" 

"EXECIO * DISKR JCL (STEM JCL. FINIS" 

DO idx=1 TO jcl.0 

   INTERPRET "QUEUE" jcl.idx 

END 

QUEUE "##" 

jc_str = "ABCDEFGHIJKLMNOPQRSTUVWXYZ0123456789" 

jcb = SUBSTR(jc_str,random(1,length(jc_str)),1) 

"SUB * END(##) JOBCHAR("jcb")" 

EXIT 

Die vorbereitete JCL, die im Beispiel aus dem Member GENER1 der Datei userid.SKELS.CNTL 
gelesen wird, sieht folgendermaßen aus: 

userid.SKELS.CNTL(GENER1) 

"//DRUCK EXEC PGM=IEBGENER" 

"//SYSUT1 DD DSN="drudat",DISP=SHR" 

"//SYSUT2 DD SYSOUT="okl 

"//SYSIN DD DUMMY" 

"//SYSPRINT DD DUMMY" 

Programm und Job-Control-Gerippe können auf diese Weise unabhängig voneinander editiert wer-
den. 

Beides lässt sich mit der dritten Variante in Verbindung bringen. Die im Stack vorbereitete Job-Con-
trol wird nicht unmittelbar durch den SUBMIT-Befehl zur Ausführung übergeben, sondern in eine 
Datei geschrieben. Dieses Vorgehen kann durchaus auch seine Vorteile haben. Zum einen könnte 
die erzeugte Ausgabedatei nochmals kontrolliert und gegebenenfalls korrigiert und ergänzt werden. 
Zum anderen bleibt der erzeugte JOB in der Datei erhalten, bis er durch eine erneute Verarbeitung 
aufs Neue erstellt wird. Im Bedarfsfall könnte er unverändert mehrmals zur Ausführung gebracht 
werden, ohne dass das REXX-Programm ablaufen muss. 

Natürlich lohnt sich dieses Vorgehen eher bei entsprechend komplexen JCL-Verarbeitungen, aber 
um zu sehen, wie das Ganze funktioniert, können wir bei unserem simplen Beispiel bleiben. 

Wir wollen für das folgende Beispiel einen kleinen Ausflug zu einem Dialog Manager Service ma-
chen, nur um zu zeigen, wie einfach und komfortabel diese Dienste in ein REXX-Programm einge-
bunden werden können, und wie viel Komfort sie dem Anwender letztlich erbringen. Nachdem die 
JCL in eine Datei geschrieben wurde, rufen wir den PDF-Editor zusammen mit den JCL-Daten auf. 
Der Anwender könnte auf diese Weise seine erzeugten JCL-Sätze sehen und im Bedarfsfall modifi-
zieren. Ob er eine wirkliche JCL-Verarbeitung wünscht, kann er selbst entscheiden, indem er den 
EDIT-Primärbefehl SUB in der Kommandozeile des Editors absetzt. Edit kann vom Anwender dann 
über die END-Funktion oder den Befehl CANCEL verlassen werden. Die Ablaufsteuerung wird sofort 
wieder an das REXX-Programm übertragen. Sind die Dialog Manager Services nicht erreichbar, wird 
stattdessen der SUBMIT-Befehl unmittelbar aus dem Programm heraus abgesetzt. 

/* REXX * JCLPRT3 *******************************/ 

DO FOREVER 

   SAY "Druckdatei vollqualifiziert (ohne Hochkomma)" 

   PARSE UPPER EXTERNAL drudat 

   IF LENGTH(drudat) = 0 THEN EXIT 

   IF SYSDSN("'"drudat"'") = "OK" THEN DO 

      DMY = LISTDSI("'"drudat"'") 

      SELECT 

         WHEN POS(SYSDSORG,"PS PO") = 0 THEN 

            msg = "Falsche Dateiorganisation" 

         WHEN SYSDSORG = "PO" & , 

            RIGHT(drudat,1) <> ")" THEN 

            msg = "PO-Datei ohne Memberangabe" 

         WHEN SYSLRECL > 132 THEN 

            msg = "Satzlänge zu groß" 

         OTHERWISE LEAVE 

      END 

   END 

   ELSE msg = drudat "nicht verfügbar" 

   SAY msg 

END 

SAY "Outputklasse (A, R oder X(=Standard))" 

PARSE UPPER EXTERNAL okl 

IF WORDPOS(okl,"A R X") = 0 THEN okl = "X" 

QUEUE "//DRUCK    EXEC PGM=IEBGENER" 

QUEUE "//SYSUT1   DD DSN="drudat",DISP=SHR" 

QUEUE "//SYSUT2   DD SYSOUT="okl 

QUEUE "//SYSIN    DD DUMMY" 

QUEUE "//SYSPRINT DD DUMMY" 

QUEUE 

IF SYSDSN(SUB.TEMP) = "OK" THEN "DEL SUB.TEMP" 

"ALLOC DD(JCL) DA(SUB.TEMP) NEW SPACE(1)" , 

   "TRACK LRECL(80) BLKSIZE(3120) DSORG(PS)" 

"EXECIO" QUEUED() "DISKW JCL (FINIS" 

IF SYSVAR(SYSISPF) = "ACTIVE" THEN , 

   ADDRESS ISPEXEC "EDIT DATASET(SUB.TEMP)" 

ELSE 

   "SUB SUB.TEMP" 

"FREE DD(JCL)" 

EXIT 

1.25. Modularisierung 

Modularisierung ist eines der Zauberworte in der Programmierung. Es ist besser, kleine, übersicht-
liche Programme zu schreiben, die gegebenenfalls als Module in anderen Verarbeitungen benutzbar 
sind. Klein ist dabei relativ. Sind für den Einsteiger 30 Programmzeilen, die fehlerfrei funktionieren 
ein gewaltiges Pensum, lacht der Profi vielleicht über Minimalprogrämmchen von lediglich einigen 
tausend Statements. Ich würde als grobe Richtung nach einer Programmgröße von mehr als 300 
Zeilen überdenken, ob durch Modularisierung nicht einiges verbessert werden kann. 

Wird in einem Programm ein Befehl oder eine Befehlsfolge mehr als einmal an unterschiedlichen 
Stellen benötigt (dies kann mit einer Programmschleife nicht gelöst werden), sollte man den Weg 
der Modularisierung gehen. Auch wenn dafür in der Praxis der Kodieraufwand gering ist (man 
schreibt die Befehle ja nur einmal und kopiert sie im Programm an die Stelle, an der man sie zum 
wiederholten Male braucht), ergibt sich ein wunderbarer Nebeneffekt: Ein Befehl, einmal falsch ko-
diert und 20 mal kopiert, bereitet jedem Programmierer große Freude bei der Fehlersuche. Wird der 
Befehl als Unterroutine aufgerufen (dies kann von jedem beliebigen Punkt des Programmes aus 

geschehen), muss er nur einmal kodiert werden und kann damit in aller Regel nicht öfter als einmal 
falsch geschrieben werden - ehrlich! 

Modularisierung bietet einige wesentliche Vorteile: 

▪ Die Programmstrukturen sind wesentlich übersichtlicher. 
▪ Der Wartungsaufwand wird erheblich reduziert. 
▪ Mehrfach verwendete Programmteile müssen nur einmal kodiert werden. 
▪ Externe Unterprogramme können von anderen Programmen genutzt werden. 
▪ Modularisierte Programme sind wesentlich änderungsfreundlicher. 

Bevor wir uns aber mit den Details auseinandersetzen, müssen wir zunächst einige Ausdrücke klä-
ren, die nicht REXX-spezifisch sind. Bei Modularisierung unterscheiden wir zwischen 

▪ einer Subroutine 
▪ einem internen Unterprogramm 
▪ einem externen Unterprogramm 
▪ einer internen Funktion und 
▪ einer externen Funktion 

Keine Angst, so kompliziert ist es nun auch wieder nicht. Beachtet man ein paar Dinge bei der Ko-
dierung, kann das Ergebnis als Subroutine, Unterprogramm oder Funktion benutzt werden. Wichtig 
ist der Einsatz des Programmes. Wir müssen entscheiden, ob es intern oder extern eingesetzt wer-
den soll, und ob uns in unserer Programmlogik die Nutzung als Unterprogramm oder als Funktion 
besser entgegenkommt. 

Um eine Entscheidung über die Form des Einsatzes zu treffen, müssen wir uns ansehen, wie ein 
REXX-Programm innerhalb eines TSO-Adressraumes abläuft. Nachdem es für ein besseres Ver-
ständnis nicht nötig ist, jedes einzelne Byte mit Vornamen zu kennen, stellen wir unsere Betrachtung 
hier etwas oberflächlich an: 

Jedes REXX-Programm läuft im TSO-Adressraum als selbständige Task und verfügt über einen 
eigenen Speicherbereich. Alle Variablen, die ein Programm bildet, sind taskbezogen. Ruft ein REXX-
Programm ein externes Unterprogramm oder eine externe Funktion auf, wird hierfür eine eigene 
Task aufgebaut. Die Variablen des aufrufenden Programmes sind im aufgerufenen Programm un-
bekannt und umgekehrt. Wird eine Task beendet (Programmende), wird der belegte Speicherbe-
reich wieder freigegeben. Alle Variablen innerhalb der Task werden aufgegeben. 

Auf den folgenden Seiten werden wir uns alle Variationen detailliert betrachten. Um nicht allzu große 
Verwirrung zu stiften, nehmen wir als Kernstück für unsere Subroutine, interne Funktion, externes 
Unterprogramm, externe Funktion, internes Unterprogramm, unternes Interprogramm, progterne In-
terfunktion, funkterne Progroutine, ... wie auch immer, einen Programmabschnitt, den wir bereits 
mehrmals eingesetzt haben. Wir wollen letzten Endes erkennen können, ob eine Datei folgende 
Kriterien erfüllt: 

▪ muss vorhanden und verfügbar sein 
▪ muss PO- oder PS-Format aufweisen 
▪ im Falle einer PO-Datei muss der Membername angegeben sein 
▪ die Satzlänge darf 132 Byte nicht überschreiten 

Wir nennen unser Gebilde #STDFILE (Standard-Datei) und gehen davon aus, dass als Subroutine 
oder internes Unterprogramm eine Variable STDFILE mit dem Wert 1 füllt, sofern die zu prüfende 
Datei  unseren Standardanforderungen genügt. Anderenfalls wird die Variable mit einer 0 gefüllt. Da 
hier nicht die Prüfung im Vordergrund steht, sondern wie Modularisierung angewandt wird, wurde 
die Prüfung aus Platzgründen durch eine etwas andere Schreibweise verkürzt. 



Inside z/OS Seite 78 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

Erfolgt der Aufruf als externes Unterprogramm, wird 1 oder 0 an das übergeordnete Programm zu-
rückgereicht und liegt dort in Form der speziellen Variablen RESULT vor. 

Wird das Ganze als Funktion eingesetzt (intern oder extern), liefert die Funktion im Ablauf den je-
weils zutreffenden Wert (0 für Nein, 1 für Ja) anstelle ihrer selbst, sodass mit der Funktion unmittel-
bare Vergleiche und Abfragen durchgeführt werden können (IF #STDFILE(datei) THEN SAY 

"alles klar"). 

1.25.1. Die Subroutine 

Durch die Instruktion CALL wird zu einer Marke verzweigt, die hinter dem eigentlichen Programm 
liegen sollte. Die Rückkehr ins Hauptprogramm wird durch die Anweisung RETURN veranlasst. 

Innerhalb einer Subroutine wird auf den gleichen Variablenbereich zugegriffen wie im Hauptpro-
gramm. Aus diesem Grund brauchen keine Maßnahmen für den Variablenaustausch zwischen 
Hauptprogramm und Subroutine ergriffen werden. Die Subroutine läuft innerhalb der gleichen Task 
des TSO-Adressraumes ab. 

Die Nutzung als Subroutine ist sinnvoll, wenn spezielle Problemlösungen benötigt werden, die in 
einer anderen Anwendung nicht gebraucht werden können und die nicht allzu groß sind. 

/* REXX * DRUCK *********************************/ 

DO FOREVER 

   SAY "Name Druckdatei (ggf. mit Hochkomma)" 

   PARSE UPPER EXTERNAL DRUDAT 

   IF LENGTH(DRUDAT) = 0 THEN EXIT 

   CALL #STDFILE 

   IF stdfile = 1 THEN LEAVE 

   ELSE SAY DRUDAT "erfüllt die Anforderungen nicht" 

END 

/* Verarbeitung */ 

EXIT 

 

#STDFILE: 

IF SYSDSN(DRUDAT) = "OK" THEN DO 

   DMY = LISTDSI(DRUDAT) 

   IF POS(SYSDSORG,"PS PO") = 0 ! , 

      SYSDSORG = "PO" & RIGHT(strip(DRUDAT,,"'"),1) >< ")" ! , 

      SYSLRECL > 132 THEN stdfile = 0 

   ELSE stdfile = 1 

END 

ELSE stdfile = 0 

RETURN 

Problematisch ist in der Subroutine die Nutzung des gleichen Variablenbereiches wie im Hauptpro-
gramm, so praktisch es einerseits auch ist. Wenn nicht sauber und aufmerksam kodiert wird, ist 
schnell mal eine Variable überschrieben, die im Hauptprogramm noch gebraucht worden wäre. Mehr 
Sicherheit bietet hier das Unterprogramm. 

1.25.2. Interne und externe Unterprogramme 

Unterprogramme verwenden nicht den gleichen Variablenpool wie die Hauptprogramme. Mit der 
Anweisung PROCEDURE werden die im Hauptprogramm definierten Variablen isoliert und erst wie-
der verfügbar, wenn durch RETURN ins Hauptprogramm zurückgekehrt wird. 

Wird das Unterprogramm beendet, wird die Task, in der es gelaufen ist, wieder freigegeben. Alle 
Variablen sind damit gelöscht. 

Sehen wir uns unsere Subroutine mal als internes Unterprogramm an: 

/* REXX * DRUCK *********************************/ 

DO FOREVER 

   SAY "Name Druckdatei (ggf. mit Hochkomma)" 

   PARSE UPPER EXTERNAL DRUDAT 

   IF LENGTH(DRUDAT) = 0 THEN EXIT 

   CALL #STDFILE DRUDAT 

   IF result = 1 THEN LEAVE 

   ELSE SAY DRUDAT "erfüllt die Anforderungen nicht" 

END 

/* Verarbeitung */ 

EXIT 

 

#STDFILE:PROCEDURE 

ARG DRUDAT 

IF SYSDSN(DRUDAT) = "OK" THEN DO 

   DMY = LISTDSI(DRUDAT) 

   IF POS(SYSDSORG,"PS PO") = 0 ! , 

      SYSDSORG = "PO" & RIGHT(strip(DRUDAT,,"'"),1) >< ")" ! , 

      SYSLRECL > 132 THEN stdfile = 0 

   ELSE  stdfile = 1 

END 

ELSE stdfile = 0 

RETURN stdfile 

Nachdem die Variablenbereiche des Haupt- und Unterprogramms voneinander abgeschottet liegen, 
müssen Informationen in beiden Richtungen hin- und hergereicht werden. Dies ist das gleiche Ver-
fahren, als läge das Unterprogramm als eigenes Member der REXX-Programmbibliothek vor (exter-
nes Unterprogramm). Der Anreiz, ein Unterprogramm intern zu schreiben liegt darin, eine generelle 
Trennung der Variablen zwischen Haupt- und Unterprogramm zu haben, einzelne Variable aber in 
beiden Richtungen durchlässig zu definieren. So hat man auf der einen Seite deutlich mehr Schutz 
vor unbeabsichtigtem Überschreiben von Variablen, andererseits muss man sich um Über-
gabe/Übernahme einzelner Variablen nicht gesondert kümmern. Diesen Service erhalten wir, indem 
das Schlüsselwort EXPOSE in die Zeile PROCEDURE kodiert wird. Variablen, die durchlässig sein 
sollen, werden hinter EXPOSE angegeben. Sehen wir uns auch hierzu ein kleines Beispiel an: 

/* REXX * DRUCK *********************************/ 

DO FOREVER 

   SAY "Name Druckdatei (ggf. mit Hochkomma)" 

   PARSE UPPER EXTERNAL DRUDAT 

   IF LENGTH(DRUDAT) = 0 THEN EXIT 

   CALL #STDFILE 

   IF result = 1 THEN LEAVE 

   ELSE SAY DRUDAT "erfüllt die Anforderungen nicht" 

END 

/* Verarbeitung */ 

EXIT 

 

#STDFILE:PROCEDURE EXPOSE DRUDAT 

IF SYSDSN(DRUDAT) = "OK" THEN DO 

   DMY = LISTDSI(DRUDAT) 

   IF POS(SYSDSORG,"PS PO") = 0 ! , 

      SYSDSORG = "PO" & RIGHT(strip(DRUDAT,,"'"),1) >< ")" ! , 

      SYSLRECL > 132 THEN stdfile = 0 

   ELSE stdfile = 1 

END 

ELSE stdfile = 0 

RETURN stdfile 

Werden keine Variablen durch EXPOSE vom Zugriffschutz ausgeklammert, werden interne und ex-
terne Unterprogramme im Hinblick auf Variablen gleich behandelt.  

Zu übergebende Variablen werden hinter dem Namen des Unterprogrammes angehängt und im 
Unterprogramm mit der Anweisung ARG entgegengenommen (Positionsparameter). Die Steuerung 
an das Hauptprogramm wird durch RETURN zurückgegeben. Werden mit der Instruktion eine oder 
mehrere Variablen zurückgereicht, liegt im Hauptprogramm der gesamte Parameterstring in RE-
SULT vor.  

Externe Unterprogramme sind alles in allem sicher die sinnvollere Lösung, wenngleich sie uns, was 
den beidseitigen Informationsaustausch angeht, oft an die Grenze zum Wahnsinn treiben. In diesem 
Fall liegt das Unterprogramm als eigenständiges Member einer REXX-Ladedatei vor. Wir betrachten 
uns in den nächsten Beispielen den Informationsaustausch zwischen zwei REXX-Programmen, die 
beide Member einer REXX-Programmbibliothek sind. Beginnen wir mit den einfachen Dingen: 

/* REXX * HP1 ***********************************/ 

a=2 

SAY a 

CALL UP1 a 

a=result 

SAY a 

EXIT 

/* REXX * UP1 ***********************************/ 

ARG zahl 

zahl = zahl * 2 

RETURN zahl 

Das Hauptprogramm übergibt einen Wert, der im Unterprogramm durch ARGument empfangen wird. 
Das Unterprogramm reicht einen Wert über die Instruktion RETURN an das Hauptprogramm zurück, 
wo er in Form der speziellen Variablen RESULT vorliegt. Nächster Versuch: 

/* REXX * HP1 ***********************************/ 

a=2 

b=3 

CALL UP1 a b 

erg=result 

SAY erg 

EXIT 

 

/* REXX * UP1 ***********************************/ 

ARG zahl1 zahl2 

zahl = zahl1 * zahl2 

RETURN zahl 

Die Übergabe mehrerer Parameter an das Unterprogramm und der dortige Empfang bereiten keine 
Probleme, solange sich eine Zuweisung auf Wortbasis durchführen lässt.  

Die Rückgabe des Ergebniswertes über RETURN ist auch vollkommen unproblematisch, solange 
das zurückzureichende Ergebnis aus nur einem Wort besteht. Generell könnten hinter RETURN 
zwar beliebig viele Variablen oder Literale angegeben werden, im Hauptprogramm gibt es aber nur 
ein RESULT. Auf ein Neues. 

/* REXX * HP1 ***********************************/ 

a=2 

b=3 

CALL UP1 a b 

SAY result 

EXIT 

 

/* REXX * UP1 ***********************************/ 

ARG zahl1 zahl2 

zahl1 = zahl1 * zahl1 

zahl2 = zahl2 * zahl3 

RETURN zahl1 zahl2 

Liegen in RESULT mehrere einzelne Werte, die im Hauptprogramm separat weiterverarbeitet wer-
den sollen, müssen sie entsprechend getrennt werden. Es könnte aber auch sein, dass ein Unter-
programm eine Reihe von Parametern zurückreicht und die Anzahl dieser Werte situationsabhängig 
ist. Hier wäre denkbar: 

/* REXX * HP1 ***********************************/ 

CALL UP1 

max_parm = WORDS(result) 

DO cnt = 1 to max_parm 

   parm.cnt = WORD(result,cnt) 

   SAY cnt". Wort aus RESULT von UP1:" parm.cnt 

END 

EXIT 

 

/* REXX * UP1 ***********************************/ 

SAY "Was haben Sie beruflich schon alles erreicht?" 

SAY "Bitte eingeben (maximal eine Bildschirmseite)" 

PARSE EXTERNAL eingabe 

RETURN eingabe 

Was auch immer das Unterprogamm an Informationen über RETURN gebracht hat (vielleicht auch 
nichts), im Hauptprogramm liegt die gesamte Eingabe als RESULT vor. MAX_PARM enthält die 
Anzahl der Worte, die über RESULT geliefert wurden. In Schleife wird ein Variablenstamm PARM. 
aufgebaut. Je eine Variable enthält der Reihe nach je ein Wort aus RESULT. Die Anzahl gebildeter 
Variablen des Stammes PARM. ist identisch mit dem Wert in MAX_PARM. Verzeiht bitte die simplen 
Beispiele. Selbstredend kommen derart einfache Dinge im täglichen Leben nicht immer auf uns zu. 
Das wirkliche Leben ist viel schlimmer. Also los: 

Wir denken uns etwas Schönes aus. Ein Programm empfängt einen Mitarbeiternamen. Ein Unter-
programm sucht diesen Namen in einer Datei und ermittelt Straße, Ort, Telefonnummer und Ge-
burtsdatum. Diese Informationen werden an das Hauptprogramm zurückgereicht. Kleinigkeit, werdet 
ihr sagen. Also legen wir noch eins drauf: Nicht alle Informationen müssen vorhanden sein. So kann 
es sein, dass zu einem Namen Straße und Ort, zu einem anderen Namen Postleitzahl und Geburts-
datum und zu einem weiteren nur Telefonnummer und Geburtsdatum ermittelt werden können. Die 
anderen Variablen sind BLANK und werden bei der späteren Worttrennung im Hauptprogramm nicht 
berücksichtigt. Wir müssen uns selbst etwas überlegen: 

/* REXX * HP1 ***********************************/ 

DO FOREVER 

   SAY "Name ?" 

   PARSE EXTERNAL name 

   IF LENGTH(name) = 0 THEN LEAVE 

   CALL UP1 name 

   PARSE VAR result str "#" ort "#" tel "#" geb 

   SAY str "/" ort "/" tel "/" geb 

END 

SAY "Bye, bye..." 

EXIT 

/* REXX * UP1 ***********************************/ 

PARSE ARG name 

"ALLOC DD(EIN) DA(........" 

"NEWSTACK" 

"EXECIO * DISKR EIN (FINIS" 

u.="" 

DO QUEUED() 

   PARSE PULL SATZ 

   IF LEFT(satz,LENGTH(name)) = name then do 

      u.str = strip(substr(satz,40,20),T) 

      u.ort = strip(substr(satz,65,30),T) 

      u.tel = strip(substr(satz,100,20),T) 

      u.geb = strip(substr(satz,125,8),T) 

      LEAVE 

   END 

END 

"DELSTACK" 

"FREE DD(EIN)" 

ergebnis = u.str"#"u.ort"#"u.tel"#"u.geb 

RETURN ergebnis 

Natürlich wären da noch einige andere Möglichkeiten. Wir könnten den Satz im Unterprogramm in 
den Stack legen und ihn dann im Hauptprogramm wieder entnehmen. Oder noch besser: Für den 
Fall, dass unser Hauptprogramm Daten im Stack hält, die es später noch benötigt, bauen wir einen 
Puffer in den Stack, legen im Unterprogramm unsere Daten in den Pufferbereich, entnehmen sie im 
Hauptprogramm wieder und lösen dort den Puffer wieder auf. 

Weiter verkomplizieren lässt sich die ganze Geschichte problemlos bis ans Ende aller Tage. Wir 
können, wenn wir die Beispiele für Datenaustausch zwischen zwei REXX-Programmen (extern) 
nochmals betrachten, zu folgendem Ergebnis kommen: 

Der Austausch einfacher Informationen innerhalb gewisser Regeln ist zwischen zwei REXX-Pro-
grammen sehr einfach. Je komplexer der Austausch von Informationen aber wird, desto schwieriger 
gestaltet sich die Kodierung. Überschreiten wir dabei eine gewisse Grenze, artet schnell in Arbeit 
aus, was als Zeitvertreib begann. Selbstverständlich haben wir in der Programmierung  "geht nicht" 
längst aus unserem Wortschatz gestrichen. Es stellt sich letztlich nie die Frage, ob ein Problem 
lösbar ist (diese Frage könnt ihr euch im Vorfeld bereits mit einem kräftigen "JA" beantworten), son-
dern immer nur wie es lösbar ist. Und wie groß ist der erforderliche Aufwand. 

Ich möchte diese Problematik hier nicht weiter vertiefen. Im wirklichen Leben ist alles anders. Wenn 
wir mit den einfachen Varianten des Informationsaustausches unter REXX nicht zufrieden sind, neh-
men wir unter TSO den Variablenservice des Dialog Managers zu Hilfe. Der erlaubt auf einem sehr 
hohen Komfortniveau den tasküberschreitenden Austausch von Informationen. Hierzu später mehr. 

Der Aufruf eines externen Unterprogrammes kann auf verschiedene Arten erfolgen. Abhängig von 
der Art des Aufrufes können wir ein geliefertes Ergebnis unterschiedlich weiterverarbeiten. Soll ein 
Programm aber wahlfrei auf beide Arten aufgerufen werden können, sollte dies bei der Erstellung 
bereits berücksichtigt werden. Das erfordert ein gewisses Maß an Disziplin bei der Kodierung. 

Im Anschluss sehen wir uns die beiden Formen des Aufrufes an und betrachten uns die Regeln, die 
für die jeweilige Form des Aufrufes einzuhalten sind. 

1.25.3. Aufruf als Unterprogramm 

call 'NAME' param1 param2 ... paramx 

• Der Programmname muss in Großbuchstaben geschrieben sein. 

• Ist der Programmname zwischen Apostrophe eingeschlossen, wird nicht im Hauptprogramm ge-
sucht. 

• Bei Datenübergabe an das Unterprogramm werden 
o Parameter durch BLANK voneinander getrennt. 
o Parameterketten werden durch KOMMA voneinander getrennt. 

• Die Datenrückgabe erfolgt über RETURN und steht im Hauptprogramm in RESULT. 

1.25.4. Aufruf als Kommando 

'NAME' param1 param2 ... paramx 

• Es können nicht nur REXX-Programme aufgerufen werden. 

• Bei Datenübergabe an das Unterprogramm sind nur Parameter, aber keine Parameterketten 
möglich. 

• Die Datenrückgabe über RETURN/RESULT ist nicht möglich. 

• In der Anweisung EXIT kann ein numerischer Wert zurückgereicht werden, der im Hauptpro-
gramm als RC empfangen wird. 

/* REXX * HP1 ***********************************/ 

"UP1" parm1 parm2 

IF rc > 0 THEN SAY "Fehler in UP!, RC:" rc 

EXIT 

 

/* REXX * UP1 ***********************************/ 

: 

IF ... THEN rcode = 0 

ELSE rcode = 8 

EXIT rcode 

1.25.5. Funktionsaufruf 

Selbstgeschriebene Funktionen werden im Allgemeinen UDF's (User Defined Functions) genannt. 
In der Klammer der Funktion können entweder keine, eine oder mehrere Variablen angegeben wer-
den, wobei im letzten Fall die Variablen durch KOMMA(,) voneinander getrennt werden. 

Prinzipiell können Funktionen intern oder extern geschrieben werden. Einige Dinge sind bei Nutzung 
von UDF´s zu beachten: 

• Bei internen Funktionen wird ein gleichnamiges Label angesprungen. 

• Bei externen Funktionen wird ein gleichnamiges Member in SYSPROC/SYSEXEC gesucht.  

• Aus Gründen der Sicherheit empfiehlt sich bei internen Funktionen das Abgrenzen der Variablen 
durch das Statement PROCEDURE.  

• Daten werden mit der Instruktion ARG entgegengenommen und über RETURN wieder zurück-
gereicht. Dieser Wert ersetzt dann  den Funktionsaufruf. 

• Werden der Funktion beim Aufruf mehrere Parameter, durch Komma getrennt, übergeben, muss 
die Funktion die Variablentrennung im ARGument-Statement ebenfalls über Komma vornehmen. 

• Die Angabe eines Ausdruckes bei RETURN ist zwingend, da ansonsten eine Fehlermeldung 
ausgegeben wird. 

Wir sehen uns am besten ein kleines Beispiel an. In einer internen Funktion wird der deutsche Wo-
chentagsname des aktuellen Datums ausgegeben. 

/** REXX ************************************/ 

say "Heute ist" tag()", der ",  

    translate("Tt.Mm.CcJj",date(s),"CcJjMmTt") 

exit 

tag: procedure 

tag.0="Montag" 

tag.1="Dienstag" 

tag.2="Mittwoch" 

tag.3="Donnerstag" 

tag.4="Freitag" 

tag.5="Samstag" 

tag.6="Sonntag" 

x=date(b) // 7 

return tag.x 

Die Bildschirmausgabe könnte folgendermaßen aussehen: 

Heute ist Freitag, der 22.04.2022 

Vielleicht habt ihr euch schon einmal gefragt, wie ein Programm die beweglichen Feiertage ermittelt. 
Letztlich basieren alle beweglichen Feiertage auf Ostern und stehen in einem festen Abstand zum 
Ostersonntag. Dieser wiederum ist abhängig vom ersten Vollmond im Frühling. Wir müssen also das 
Datum des Ostersonntages berechnen und alle unsere Feiertagsprobleme sind gelöst (einige löst 
vielleicht auch die Regierung durch entsprechende Streichungen). 

Wir wollen das Datum des Ostersonntags über eine externe Funktion ermitteln. Die Funktion sollte, 
sofern sie keinen Übergabeparameter enthält, den Ostersonntag des aktuellen Jahres errechnen. 
Anderenfalls kann innerhalb der Klammern der Funktion ein beliebiges Jahr (vierstellig) angegeben 
werden. Auf Plausibilitätsprüfungen innerhalb der Funktion wird verzichtet. 

/* REXX * HP1 ***********************************/ 

SAY "Ostersonntag ist dieses Jahr der" OSO() "und" 

SAY "nächstes Jahr der" OSO(RIGHT(DATE(),4)+1) 

EXIT 

 

/* REXX * FUNKTION OSO **************************/ 

ARG jahr 

IF LENGTH(jahr) = 0 THEN jahr = RIGHT(DATE(),4) 

a = jahr // 19 

b = jahr // 4 

c = jahr // 7 

d = (19 * a + 24) // 30 

e = ((2 * b) + (4 * c) + (6 * d) + 5) // 7 

f = 22 + d + e 

IF f <= 31 then mm = "03" 

ELSE DO 

   IF f = 57 THEN f = 50 

   IF f = 56 & d = 28 & e = 6 & a > 10 THEN f = 49 

   mm = "04" 

   f = f - 31 

END 

RETURN f"."mm"."jahr 

Es ist nicht schwierig, so eine Formel zu schreiben. Herr Gauß hat es auch geschafft. 

1.25.6. Default-Übergabewerte bei Function-Calls 

Es wird oft gefragt, ob es analog zu CLISTen (und dort im PROC-Statement) die Möglichkeit gibt, 
bei Aufruf von Unterroutinen für Fehlende Übergabeparameter Defaultwerte zu nutzen. 

Die simple Antwort ist NEIN. Aber geht nicht, gibt's nicht. Wird die Subroutine als Function aufgeru-
fen, würde man dies zustande bringen (siehe unter "Parse für die Ausgebufften"). 

1.26. REXX im Batch 

Der grundlegende Vorteil, ein REXX-Programm im Hintergrund ablaufen zu lassen, ist, dass das 
Terminal nach dem Absetzen des Programmstarts sofort wieder zur Verfügung steht. Zeitintensive 
Prozesse können im Batch besser verarbeitet werden als in einem LOGON-Adressraum. In vielen 
Unternehmen sind die Ablaufprioritäten einer LOGON-Session so geregelt, dass mit zunehmender 
CPU-Zeit die Ablaufpriorität abnimmt (je länger eine REXX läuft, desto langsamer wird sie). 

Verarbeitungen, die ein gewisses Maß an Komplexität überschreiten, oder eine große Datenmenge 
zu bewältigen haben, müssen nicht unbedingt mit REXX realisiert werden. Es soll daneben auch 
noch andere Werkzeuge geben (behaupten zumindest einige Leute immer wieder). 

Die große Stärke von REXX liegt in der komfortablen Kodierung kleinerer Anwendungen, dem Ein-
satz als leistungsstarke Trägersprache für Dialog Manager Service oder EDIT-Makros. Dennoch 
wollen wir hier Wege aufzeigen, was man unternehmen kann, wenn die Laufzeit eines REXX-Pro-
grammes innerhalb einer LOGON-Session zu wünschen übrig lässt.  

Läuft ein REXX-Programm nicht im Foreground, sollte gewährleistet sein, dass das Programm auch 
ohne Bedienereingaben am Terminal auskommt. 

Grundsätzlich müssen wir unterscheiden, ob wir in einem Batch-Adressraum TSO starten und 
REXX-Programme dort ablaufen sollen, oder ob REXX in einem Nicht-TSO-Adressraum zum Ein-
satz kommt. Zuweisungen müssen entsprechend über Job-Control-Statements erfolgen. 

1.26.1. REXX im Batch in einem TSO-Adressraum 

Diese Variante ist die gängigste, wenn es schon sein muss. Über das Programm IKJEFT01 wird die 
TSO-Umgebung in einem Batch-Adressraum aufgebaut, von der aus CLISTen oder REXX-Pro-
gramme aufgerufen, oder TSO-Kommandos zur Ausführung gebracht werden können. Gehen wir 
davon aus, dass über ein REXX-Programm eine sehr große Datei bearbeitet werden soll. Diese 
Verarbeitung würde das Terminal über einen zu langen Zeitraum blockieren und die Ablaufpriorität 
der LOGON-Session soweit reduzieren, dass die einzig sinnvolle Aktion in einem erneuten LOGON 
liegen würde. 

Das nachfolgende Job-Control-Beispiel zeigt, wie der Ablauf des REXX-Programmes im Hintergrund 
aussehen könnte: 

//SP11 JOB (11199002),'WITTEMANN',NOTIFY=SP1 

//TSO  EXEC PGM=IKJEFT01,DYNAMNBR=30,REGION=4096K 

//SYSEXEC  DD SP1.MVS.EXEC,DISP=SHR 

//         DD SP1.FUNCTION.EXEC,DISP=SHR 

//SYSTSPRT DD SYSOUT=X 

//SYSTSIN  DD * 

  %UPDATE1 SP1.TEST.DATA 

/* 

Durch den Aufruf von IKJEFT01 haben wir TSO im Batch-Adressraum. Damit die Startprozedur UP-
DATE1 impliziert aufgerufen werden kann, müssen wir für die entsprechenden Zuweisungen sorgen. 
Die DD-Anweisung SYSEXEC weist zwei Dateien für REXX-Programmaufrufe zu. Damit sie aber 
auch wirklich durchsucht werden, ist als erste Aktion im aufgebauten TSO-Batch-Adressraum der 
Aufruf des Dienstprogrammes EXECUTIL mit dem Parameter SEARCHDD(YES). Er bewerkstelligt 
die Durchsuchung der beiden unter SYSEXEC zugewiesenen Dateien bei REXX-Programmaufru-
fen. 

1.26.2. REXX im Batch in einem NON-TSO-Adressraum 

Auch hier macht es keinen Sinn in die letzten Tiefen vorzudringen. Ein Beispiel wollen wir uns aber 
ansehen. Nachdem uns in einem NON-TSO-Adressraum implizierte Programmaufrufe für REXX 
nicht möglich sind, und der EXEC-Befehl auch nicht zur Verfügung steht (ist ja ein TSO-Kommando), 
müssen wir einen anderen Weg gehen. 

Versuchen wir es unter Einbeziehung des Programmes IRXJCL, das wir über Job-Control starten 
und das seinerseits zum Ablauf eines REXX-Programmes führen wird. Unter IRXJCL wird das State-
ment SYSEXEC als Standardsymbol für REXX-Aufrufe angenommen. Der Ablauf ist dem Verfahren 
in einem TSO-Batch-Adressraum ähnlich. Der gravierendste Unterschied besteht darin, dass keine 
TSO-Kommandos, TSO-REXX-Kommandos und auch keine erweiterten TSO-Funktionen zur Ver-
fügung stehen. Der Anwendungsbereich engt sich dadurch ganz von alleine drastisch ein, da eine 
der erwähnten Komponenten in vielen REXX-Programmen vorkommt. 

Sehen wir uns in Folge die erforderliche Job-Control, das REXX-Programm, dessen Output und das 
erzeugte Job-Log-Protokoll der JCL-Verarbeitung zunächst kommentarlos an. 

Job-Control für REXX im MVS-Adressraum 

//SP11 JOB (4711),'WITTEMANN',NOTIFY=SP1 

//MVSRX EXEC PGM=IRXJCL,PARM='RXMVS01 Hallo Looser' 

//SYSEXEC  DD DSN=SP1.MVS.EXEC,DISP=SHR 

//SYSTSPRT DD DSN=SP1.RXMVS.LOG,DISP=OLD 

//SYSTSIN  DD * 

August Nieselprim 

Jeremias Schrumpelhut 

/* 

Quellencode des REXX-Programmes 

/* REXX * RXMVS01 *****************************/ 

SAY "Startzeit:" TIME() 

PARSE ARG parm1 parm2 

SAY "Empfangene Parameter:" parm1 parm2 

DO UNTIL LENGTH(jcl_data) = 0 

   PARSE PULL jcl_data 

   SAY jcl_data 

END 

SAY "Endezeit:" TIME() 

EXIT 125 

Ausgabe des REXX-Programmes (Dataset SP1.RXMVS.LOG) 

Startzeit: 14:00:03 

Empfangene Parameter: Hallo Looser 

August Nieselprim 

Jeremias Schrumpelhut 

Endezeit: 14:00:05 

JES-JobLog der Verarbeitung (Auszug) 

ALLOC FOR SP11 MVSRX 

6C7 ALLOCATED TO SYSEXEC 

854 ALLOCATED TO SYSTSPRT 

JES2 ALLOCATED TO SYSTSIN 

SP11 MVSRX - STEP WAS EXECUTED - COND CODE 0125 

: 

Die Job-Control 

Über das Programm IRXJCL kann REXX in einem NON-TSO-Adressraum ablaufen. SYSEXEC ist 
für das Programm das standardgemäße Statement für REXX-Programmausführung. Als REXX-Pro-
grammbibliothek wird die Datei SP1.MVS.EXEC zugewiesen. 

Die Parameterangaben an das Programm IRXJCL werden folgendermaßen interpretiert: Der erste 
Parameter wird als der Name des zu startenden REXX-Programmes übernommen. Alle weiteren 
Parameter werden als Übergabewerte für das REXX-Programm interpretiert. Diese Informationen 
kann das REXX-Programm über die Instruktion ARG aufnehmen. 

Für die SAY-Ausgaben des REXX-Programmes wird über den symbolischen Namen SYSTSPRT 
die Datei SP1.RXMVS.LOG zugewiesen. 

Daten, die über das Statement SYSTSIN mit der Job-Control an den INTERNAL READER überge-
ben werden, können im laufenden REXX-Programm wie vom Stackbereich gelesen werden. 

Das REXX-Programm 

Es übernimmt über die Instruktion ARG die beiden Parameter "Hallo" und "Looser" aus der Job-
Control. Über die UNTIL-Schleife werden durch die PARSE-Anweisung der Reihe nach alle Datens-
ätze gelesen, die innerhalb der Job-Control durch das Statement SYSTSIN in die Spool-Eingabebe-
reiche gelesen wurden, in der Variablen JCL_DATA abgelegt und durch SAY ausgegeben. 

Zu Beginn und Ende des Programmes wird die Systemzeit ausgegeben. 

Bei Ende des Programmes wird als Return Code 125 zurückgereicht. 

Die Ausgabedatei (SP1.RXMVS.LOG) 

Sie enthält der Reihe nach Satz für Satz alle Ausgaben, die im REXX-Programm über die Instruktion 
SAY erzeugt wurden. 

Das JOB-LOG-Protokoll 

Hier sehen wir auszugsweise nur die wichtigsten Informationen. Die Initiator-Zuweisungen lassen 
die physischen Geräteadressen der Platten erkennen, auf der die Dateien SP1.MVS.EXEC ('6C7'x) 
und SP1.RXMVS.LOG ('854'x) liegen. Die Zuweisung für Dateneingabe (SYSTSIN) erfolgt auf JES2. 
Der bei Programmende gelieferte Return Code des REXX-Programmes war 125. 

1.26.3. REXX im Batch als Instream-Daten 

Ab und an sehr praktisch ist, wenn eine REXX im Batch direkt als Instream-Daten geliefert werden 
kann. Hierzu ein kleines Beispiel: 

//********************************************************************* 

//* This JCL demonstrates how to create and execute an in-stream 

//* REXX procedure. The PARM field contains only a single x'00'. 

//********************************************************************* 

//IRXJCL    EXEC  PGM=IRXJCL,PARM=' ' 

//SYSTSPRT  DD  SYSOUT=* 

//SYSTSIN   DD  DUMMY 

//SYSEXEC   DD  * 

say "Hello world! Date = "date()", time = "time() 

EXIT 

/* 

//* 

Nochmal: A-C-H-T-U-N-G: Die Angabe bei PARM ist ein „00“x, kein Blank! Es sieht nur so aus. 

Natürlich gibt es noch weitere grandiose Ideen, wie man so etwas bewerkstelligen kann. Anderer-
seits: Man muss nicht alles machen, nur weil es geht. 

1.27. SysCalls 

SysCalls bilden die Schnittstelle zu den Unix System Services. Da wir uns schwerpunktmäßen mit 
TSO auseinandersetzen wollen, sei hier für Details auf des IBM-Manual "z/OS Using REXX and 
z/OS UNIX System Services" verwiesen. Dennoch sollen ein paar Beispiele zeigen, wie das Codeing 
aussehen könnte. Die Beispiele sind im Wesentlichen dem o.g. Manual entnommen. 

1.27.1. Lesen und anzeigen von ROOT 

/* REXX 

*/ 

Call syscalls "ON" 

Address syscall "READDIR /root." 

Do §=1 to root.0 

   Say root.§ 

End 

exit 

1.27.2. Zeigt das aktuelle Directory und listet ein beliebiges 

/* REXX 

*/ 

parse arg dir 

call syscalls "ON" 

address syscall "getcwd cwd" 

say "Aktuelles Directory:" cwd 

if dir = "" then exit 

"ls -l dir" 

Return 

1.27.3. Informationen über das gemountete Filesystem 

/* REXX 

*/ 

call syscalls "ON" 

address syscall 

numeric digits 12 

"getmntent m." 

do §=1 to m.0 

   "statfs" m.mnte_fsname.i "s." 

   j=s.stfs_avail * s.stfs_blocksize 

      if length(m.mnte_path.i)>20 then , 

         say m.mnte_path.i strip(m.mnte_fsname.i) "("j")" 

      else , 

         say left(m.mnte_path.i,20) strip(m.mnte_fsname.i) "("j")" 

end 

exit 

1.28. zOS UNIX REXX Funktionen 

Es gibt eine Reihe von Unix-Funktionen, welche auch einem REXX-Programm unter TSO zur Ver-
fügung stehen. Einige von denen sind besser als ihre Pendants im TSO. Eine umfassendere Be-
schreibung ist im Band 2 Referenz zu finden. Ich bitte aber gleich hier um Nachsicht, dass nicht alles 
beschrieben wird. Nicht umsonst hat das IBM-Manual zu diesem Thema knappe 300 Seiten. 

Hier ein kurzer Überblick 

1.28.1. I/O-Funktionen 

Diese Funktionen sind die Alternative zu ALLOC/EXECIO. über die Unix I/O Funktionen kann Datei-
verarbeitung wie auf anderen Plattformen (z.B.: Windows und Regina REXX) erfolgen. Früher 
konnte man auch das sogenannte Stream-Function-Package bei der IBM ordern. 

Wir wollen hier nicht alles berücksichtigen, was an dieser Stelle möglich ist, sondern nur diejenigen 
Funktionen, welche für Dateiverarbeitung von Interesse sind. 

1.28.1.1. Fluchtsequenzen 

Spezielle Variablen können als Steuerzeichen benutzt werden. Hier ein kleiner Auszug. Details sind 
im Manual "Using REXX and Unix System Services". 

ESC_A "2F"x Signalton 

ESC_B "16"x Backspace 

ESC_N "15"x Newline 

ESC_R "0D"x Carriage Return 

1.28.1.2. CHARS() und LINES() 

Lassen erkennen, ob in Einer Datei noch lesbare Daten vorhanden sind. CHARS() tut dies auf Byte-
basis, LINES() auf der Basis von Records 

1.28.1.3. CHARIN() und LINEIN() 

Lesen von Daten auf Byteebene oder von Datensätzen 

1.28.1.4. CHAROUT() oder LINEOUT() 

Schreiben in eine Datei. Auch hier kann die Operation auf Byte- oder Satzebene erfolgen. 

1.28.1.5. STREAM() 

Explizites Öffnen und Schließen einer Datei 

1.28.1.6. Schreiben einer Datei (aus der Unix-Umgebung) 

Das Programm muss aus der Unix Umgebung gestartet werden (z.B.: ISPF/PDF Option 3.17) 

/* REXX * I/O im Unix-Umfeld * Beispiel Schreiben */ 

call build_stem 

file=stream('/home/sp1/unix_io_demo1.txt','c','open write') 

do §=1 to _ 

   rc=lineout(file,o.§) 

end 

file=stream(file,'c','close') 

say "Toll" 

exit 0 

 

build_stem: 

   _=0 

   _=_+1;o._="Der Pferd hat vier Bein," 

   _=_+1;o._="an jedem Eck ein." 

   _=_+1;o._="Wenn Pferd drei Bein hätt," 

   _=_+1;o._="umfall'n tät." 

Return 

1.28.1.7. Lesen einer Datei (aus der Unix-Umgebung) 

Das Programm muss aus der Unix Umgebung gestartet werden (z.B.: ISPF/PDF Option 3.17) 

/* REXX * I/O im Unix-Umfeld 

   Beispiel Lesen 

*/ 

file=stream('/home/sp1/unix_io_demo1.txt','c','open read')  

do while lines(file) > 0 

   data = linein(file) 

   say data 

end 

file=stream(file,'c','close') 

exit 0 

1.28.1.8. Programmstart aus der TSO-Umgebung 

Soll der Aufruf aus dem TSO erfolgen, muss dafür eine Unix-Umgebung aufgebaut werden: 

/* REXX * I/O im Unix-Umfeld 

   Beispiel Lesen aus TSO-Umgebung 

*/ 

cmd ='/home/sp1/rexx/unixio#2' 

rc=bpxwunix(cmd,,out.,err.) 

if rc=0 then do 

   do §=1 to out.0 

      say out.§ 

   end 

end 

else do 

   say "'"cmd"' with error" 

   do §=1 to err.0 

      say err.§ 

   end 

end 

exit 0 

1.28.2. BPXWUNIX 

Erlaubt die Ausführung von Shell Commands. Auf diesem Weg ist es beispielsweise möglich einen 
Stem über den SORT-Befehl auch "in place" zu sortieren. Die möglichen Parameterangaben für den 
Sort sind sehr umfangreich. Hierzu ein kleines Beispiel: 

/* REXX * Sortieren eines Stems via BPXWUNIX  

   Gefunden bei John McKown (REXX-TSO Group)  

*/ 

_=0 

_=_+1;A._="Katze" 

_=_+1;A._="Salamander" 

_=_+1;A._="Hund" 

_=_+1;A._="Vogel" 

_=_+1;A._="Maus" 

_=_+1;A._="Pferd" 

_=_+1;A._="Aligator" 

_=_+1;A._="Krokodil" 

A.0=_ 

call bpxwunix "/bin/sort",a.,b. 

say;say left("Stemsort CHAR",50,"-") "Elemente:" a.0;say 

say "Nr. Unsorted (Stem A.)-- Sorted (Stem B.)----" 

do  i=1 to a.0 

    say right(i,3) left(a.i,20) b.i 

end 

exit 

Generell lassen sich auf diese Weise alle Unix-Befehle absetzen. Die Grenzen des machbaren lie-
gen ausschließlich in der Phantasie des Entwicklers. 

1.28.3. BPXWDYN 

Ermöglich die Verbindung zu SVC99 (Dynamic Allocation). Mit BPXWDYN können gravierende Män-
gel von TSO ALLOC und FREE ausgeglichen werden. Die Nachteile der TSO-Befehle sind: 

▪ Außerhalb von TSO und IDCAMS nicht benutzbar 
▪ Keine Zuweisung von GDGs über die relative Generationsnummer möglich 
▪ Keine simple Art der Dateiverkettung möglich 
▪ Der SUBSYS-Parameter wird nicht unterstützt 
▪ Keine einfache Möglichkeit dynamische DD-Namen zuzuweisen. 

 
Bei Zuweisung einer temporären Datei sind weder DD-Name noch DS-Name nötig. Wer im An-
schluss an den Alloc aber DD-/DS-Name benötigt, kann sich beides über zu bestimmende Variablen 
liefern lassen. 

Der Aufruf von BPXWDYN kann wahlfrei über CALL oder als FUNCTION erfolgen: 

Call bpxwdyn "parameter" 

RC = bpxwdyn("parameter") 

Ich würde nicht so weit gehen, in alten Programmen den TSO-Alloc durch den BPXWDYN-Alloc zu 
ersetzen, aber bei neuen Anwendungen würde ich auf jeden Fall dem Unix-Derivat den Vorzug ge-
ben. 

Achtung: Angaben von Dateinamen müssen mit BPXWDYN immer vollqualifiziert aber ohne Apo-
strophe kodiert werden. 

In Anschluss werden die Funktionalitäten des BPXWDYN aufgezeigt: 

1.28.3.1. ALLOC 

Dynamische Zuweisung von Dateien. Sehr häufig gibt es Bedarf für temporäre Dateien, für welche 
es unbedeutend ist, wie sie heißen oder unter welchen DD-Namen sie zugewiesen werden. Benutzt 
man einen DD-Namen, der bereits existiert, wird der ALLOC nicht funktionieren. Erfolgt im Vorfeld 
ein FREE oder wird der ALLOC mit dem Parameter REUSE kodiert, wird das betreffende DD-State-
ment überschrieben. Kann immer dumm ausgehen. 

Über die Unix-Variante kann dies nicht passieren. Weder die Angabe eines DD-Namens ist nötig, 
noch die eines Dateinamens. Beides kann aber ermittelt werden, wenn es im Programm benötigt 
wird. Dabei ist sichergestellt, dass es sich sowohl beim DD-Namen als auch beim Dateinamen um 
UNIQUE Werte handelt. 

/* REXX 

*/ 

call bpxwdyn "alloc lrecl(80) space(1) tracks rtddn(dd) rtdsn(ds)" 

say "DD-Name:" dd 

say "DS-Name:" ds 

: 

Exit 

Hier ein Beispiel für den Umgang mit temporären Dateien: 

/* REXX * YBPXWDY3 * Alloc a temp ps by Yves Colliard 

*/  

call bpxwdyn"alloc space(15,15) new delete lrecl(80)", 

    "recfm(f,b) msg(ymsg1.) rtddn(ydd1)rtdsn(ydsn1)"  

if result<>0 then do  

   say "Result:" result  

      if datatype(ymsg1.0)="NUM" then do  

      do i=1 to ymsg1.0  

         say ymsg1.i  

      end  

   end  

   exit  

end 

say ydd1 ydsn1 

call bpxwdyn "free fi("ydd1")" 

exit 

1.28.3.2. CONCAT 

Dynamische Erweiterung einer Dateiverknüpfung. Für die Verkettung von Dateinamen kann eine 
Liste von DD-Namen benutzt werden, ohne dass die Dateien, welche sich dahinter verbergen explizit 
in Erfahrung gebracht werden müssen. All jene, welche derartigen über die Auswertung des TSO-
Befehls LISTALC bereits hinter sich gebracht haben, werden diese Form zu schätzen wissen. 

/* REXX YBPXWDY4 * Concat a dataset to SYSPROC / SYSEXEC  

                   by Yves Colliard 

*/ 

mydsn=arg(1)  

if mydsn="" then do  

   parse source . . . . mydsn .  

   mydsn="'"strip(mydsn)"'"  

end  

if sysdsn(mydsn)<>"OK" then do  

   say "Error give dataset name" mydsn sysdsn(mydsn) 

   exit 8  

end 

/* */  

mydsn=strip(mydsn,"B","'")  

/* */  

if bpxwdyn("alloc da('"mydsn"') shr rtddn(mydd)")=0 then do  

   if bpxwdyn("concat ddlist(sysproc,"mydd")")>0 then do  

      if bpxwdyn("concat ddlist(sysexec,"mydd")")>0 then do  

         say "Error:"  

         say "Dataset" mydsn "not concatenated to SYSPROC/SYSEXEC"  

         exit 8 

      end  

      else do  

         zedsmsg="" zedlmsg="Dataset" mydsn "concatenated to SYSEXEC" 

         ADDRESS ISPEXEC "SETMSG MSG(ISRZ001)"  

      end  
   end  

   else do  

      zedsmsg="" zedlmsg="Dataset" mydsn "concatenated to SYSPROC" 

      ADDRESS ISPEXEC "SETMSG MSG(ISRZ001)"  

   end  

end  

else do  

   say "Error:" say "Dataset" mydsn "not allocated"  

   exit 8  

end 

: 

1.28.3.3. FREE 

Dynamische Freigabe eines DD-Namens. Auch hier gibt es ein deutliches Plus gegenüber dem 
TSO-FREE. So kann beispielsweise die DISP-Angabe auch hier erfolgen. 

1.28.3.4. INFO 

Liefert eine Reihe von ALLOC-Informationen. Hierzu ein kleines Beispiel aus der Feder von Yves 
Colliard: 

/* REXX YBPXWDY1  Copyright YCOS Yves Colliard Software 2011-2011 */ 

/*----------------------------------------------------------------*/ 

/* Sample of usage of BPXWDYN - Interpret return code             */ 

/*----------------------------------------------------------------*/ 

/* Sample                                                         */ 

/* BPXWDYN - Sample                                               */ 

/*================================================================*/ 

/* TRACE ?R  */ 

/* einschalten des interaktiven Trace Modus */ 

/* */ 

trace ?r 

Numeric digits 12 

/* Sample getting infos about a DD Statement */ 

/* */ 

/* ISPPROF DD Statement allocated            */ 

call bpxwdyn "info fi(ISPPROF) inrtdsn(dsnvar) msg(ym.)" 

yres=result;call out 

/* YSPPROF DD Statement does not exist       */ 

call bpxwdyn "info fi(YSPPROF) inrtdsn(dsnvar) msg(ym.)" 

yres=result;call out 

/* DD Statement missing                      */ 

call bpxwdyn "info fi() inrtdsn(dsnvar) msg(ym.)" 

yres=result;call out 

/* Wrong parameter                           */ 

call bpxwdyn "xyz  fi() inrtdsn(dsnvar) msg(ym.)" 

yres=result;call out 

/* Missing parameter                         */ 

call bpxwdyn "info fi(yves) inrtdsn() msg(ym.)" 

yres=result;call out 

/* Wrong optional parameter                  */ 

call bpxwdyn "info fi(yves) inrtdsn(retds) msg(ym.) nix" 

yres=result;call out 

exit 

/* Analyze the return code                   */ 

out: 

say "" 

select 

when yres=0 then do 

  say "Dataset" dsnvar "found" 

end 

when yres=20 then do 

  say yres "- Invalid parameter list" 

end 

when yres>-10000 & yres<-20 then do 

  say yres "- parameter "right(yres,2)-20" in error" 

end 

when yres<-9999 & yres>-1610612737 then do 

  say yres "- Message processing error. IEFDB476 return code:", 

      right(yres,2) 

end 

otherwise 

  say yres "- dynalloc error" d2x(yres,8) 

  /* 0438 ddname not found */ 

End 

if datatype(ym.0)="NUM" then do 

  do i=1 to ym.0 

    say ym.i 

  end 

  drop ym. 

End 

return 

1.28.3.5. OUTDES 

Erstellt einen Output-Descriptor. Darauf wird hier nicht eingegangen. 

1.28.3.6. Return Codes 

Die Return Codes von BPXWDYN sind sehr vielfältig. Meist ist nur das Problem, sie richtig zu inter-
pretieren. Folgende Return Codes sind möglich: 

Return Codes 

0 Normales Ende 
20 Parameter Error 

-21 bis -9999 Parameter Schlüsselwort Error (1 bis nnn) 
-100## Error IEFDB476 hat Code ## zurückgemeldet 

>0 oder -1610612737 bis -2147483648 Fehler in Dynalloc Output Descriptor 

Beispiel 

/* REXX YBPXWDY0 * RC-Demo 

   Copyright YCOS Yves Colliard Software 2011-2011 

*/ 

Numeric digits 12 

parm = "nothing" 

call bpxwdynparm  

call out result  

 

parm = "alloc fi(hugo) dtaset(my.dataset) shr" 

call bpxwdynparm 

call out result 

 

parm = "alloc fi(hugo) da(sys1.linklib) old" 

call bpxwdynparm  

call out result /* 2100000 -Requested data set unavailable. */ 

                /* The data set is allocated to another job and */ 

                /* its usage attribute conflicts with this request. */ 

                /* Corresponding Message: IKJ56225I */ 

 

parm = "alloc fi(hugo) da(sys1.linklib) old msg(ym.)" 

call bpxwdynparm  

call out result /* 2100000 -Requested data set unavailable. */ 

                /* The data set is allocated to another job and */ 

                /* its usage attribute conflicts with this request. */ 
                /* Corresponding Message: IKJ56225I */ 

       /* IKJ56225I DATA SET SYS1.LINKLIB ALREADY IN USE, TRY LATER+ */ 

       /* IKJ56225I DATA SET IS ALLOCATED TO ANOTHER JOB OR USER */ 

Exit 

 

/* Analyze the return code */ 

 out: say ""  

say "BPXWDYN" parm  

yres=arg(1)  

select  

   when yres=0 then do  

      say "Allocation '"parm"' Ok"  

   end  

   when yres=20 then do  

      say yres "-Invalid parameter list"  

   end  

   when yres>-10000 & yres<-20 then do  

      say yres "-parameter "right(yres,2)-20" in error", 

         word(parm,right(yres,2)-20)  

   end  

   when yres<-9999 & yres>-1610612737 then do  

      say yres"-Message processing error. IEFDB476 return code:", 

         right(yres,2)  

   end  

   otherwise do 

      say yres "-dynalloc error" d2x(yres,8)  

      /* 0210 Requested data set unavailable */  

      /* 0438 ddname not found */  

   end 

end  
if datatype(ym.0)="NUM" then do  

   do i=1 to ym.0  

      say ym.i  

   end  

   drop ym.  

end  

say "" 

return 

 

Weitere Informationen sind in "Using REXX and z/OS UNIX System Services" nachzulesen. 

1.29. Rexx native Beispiele 

Eines der wichtigsten Programme und auch eines der kürzesten. Wir aktivieren den Ablaufverfolger 
und lassen einen Haltepunkt setzen. Am Haltepunkt können wir beliebige Eingaben durchführen, 
ohne im Programmablauf weiter fortzuschreiten. Auf diese Weise lassen sich alle Befehle und Funk-
tionen spielerisch erfahren. Ist beispielsweise nicht klar, wie sich eine Funktion äußern wird, kann 
sie am Haltepunkt mit einen SAY-Befehl getippt werden: 

1.29.1. Trace 

/* REXX * TRACE1  

*/ 

trace "?R" 

nop 

exit 

Etwas ähnliches könnten wir uns in der Form basteln, dass in einer Programmschleife Eingaben 
möglich sind, und diese vom Interpreter als Befehle umgesetzt werden: 

/* REXX * TRACE2  

*/ 

do forever 

   say "Befehlstest (ENTER = Ende)" 

   parse external cmd 

   if length(cmd) = 0 then leave 

   interpret cmd 

end 

say "Bye, bye..." 

exit 

Und so sähe es aus, wenn es vernünftig codiert ist: 

/* REXX * TRACE3 

*/ 

parse source . . RxName . 

address ispexec "control errors return" 

 

back_2_work: 

signal on syntax  name fehler 

signal on error   name fehler 

signal on failure name fehler 

do forever 

   say "Environment:" address() ">> Command? (EXIT => END)" 

   parse external cmd 

   if translate(cmd) = "EXIT" then leave 

   interpret cmd 

end 

say;say "Thank you for gambling with TRACE..." 

exit 

 

fehler: 

   cond_C = condition("C") 

   cond_D = condition("D") 

   txt = "Condition" cond_C 

   if cond_C = "SYNTAX" then txt = txt ">>" errortext(rc) 

   else txt = txt "in" cond_D ">> RC:" RC "bei" address()"-Command" 

   say txt;say 

   signal back_2_work 

exit 

1.29.2. Papierfalten 

In jungen Jahren, so um die Zeit der Brotvermehrung, habe ich des Öfteren mal ein kostenloses Bier 
getrunken, weil kaum jemand glaubte, dass ein Blatt Papier beliebiger Größe und Stärke nicht öfter 
als siebenmal deckungsgleich zusammengefaltet werden kann, selbst wenn man dazu einen Bogen 
Seidenpapier in der Größe eines Fußballfeldes nehmen würde. Der Biegewiderstand wird spätes-
tens beim achten Mal so groß, das ein kantengerades Aufeinanderlegen nicht möglich ist. Pro-
grammtechnisch kann das Zusammenfalten natürlich beliebig oft durchgeführt werden. Wir wollen 
über ein Programm vom Anwender bestimmen lassen, wie oft ein Blatt Papier zusammengefaltet 
werden soll und unterstellen eine Papierstärke von 0,1 mm. Nach dem ersten Falten ergeben sich 
zwei Lagen, nach dem zweiten Mal 4, dann 8, 16 und so weiter. 16 Lagen zu 0,1 mm ergeben eine 
Stapelhöhe von 1,6 mm. Das Programm soll nun abhängig von der Anzahl der Faltungen die Sta-
pelhöhe berechnen. Abhängig vom Ergebnis soll die Höhe in Millimetern, Zentimetern, Metern oder 
Kilometern ausgegeben werden. Achtung: Die Wechsel vollziehen sich deutlich schneller, als man 
glauben möchte: bei acht Faltungen wechseln wir zu Zentimetern, bei 14 zu Metern und bereits bei 
24 zu Kilometern. 

/* REXX * FALTEN  

*/ 

say "Anzahl Faltungen" 

pull anz 

h=2 ** anz * 0.1 

select 

   when h < 10  then erg = h "mm" 

   when h < 1e3 then erg = h/10 "cm" 

   when h < 1E6 then erg = h/1e3 "m" 

   otherwise erg = h/1e6 "km" 

end 

say "Stapelhöhe nach" anz "Faltungen:" erg 

exit 

1.29.3. Knobeln 

Wir bauen uns ein Knobelspiel. Wer kennt es nicht aus seiner Kindheit. Stein - Schere - Papier - 
Brunnen. Mit der Hand wird auf DREI eines der Symbole gezeigt. Die Auflösung des Spieles ge-
schieht nach folgender Regel: 

Spielobjekt Gewinnt/Verliert 

Stein Gewinnt gegen SCHERE 
Verliert gegen PAPIER und BRUNNEN 

Schere Gewinnt gegen PAPIER 
Verliert gegen STEIN und BRUNNEN 

Papier Gewinnt gegen BRUNNEN und STEIN 
Verliert gegen SCHER 

Brunnen Gewinnt gegen STEIN und SCHERE 
Verliert gegen PAPIER 

 

Nachdem wir eine MVS-Anlage nur unter großer Mühe dazu bewegen können, ihre Hand zum Brun-
nen zu formen, versuchen wir das Ganze über entsprechende Zahlen. Der Spieler tippt seine Wahl 
am Terminal ein, der Rechner wählt über einen Zufallsgenerator: 

/* REXX * KNOBELN  

*/ 

obj.1="Stein" 

obj.2="Schere" 

obj.3="Papier"; 

obj.4="Brunnen" 

erg.1="unentschieden" 

erg.2="ich habe gewonnen" 

erg.3="Du hast gewonnen" 

do forever 

   say center("< Heinzi's kleine Knobelei >",79,'-') 

   say;say;say;say 

   do i=1 to 4 

      say center(I "=" left(obj.i,10),79) 

   END 

say 

say center("E = E N D E",79);say;say;say 

say center("Triff Deine Auswahl",79);say;say 

do until s >= 1 & s <= 4 ! s = "E" 

   say "Eingabe?" 

   parse upper external s 

end 

if s = "E" then leave 

s=trunc(s) 

r=random(1,4) 

say;say;say "Du hast" obj.s", ich habe" obj.r 

select 

   when r=s then e=1 

   when r=1 then if s=2 then e=2 

                        else e=3 

   when r=2 then if s=3 then e=2 

                        else e=3 

   when r=3 then if s=2 then e=3 

                        else e=2 

   otherwise 

      if s=3 then e=3 

      else e=2 

end 

say erg.e "- weiter mit ENTER" 

pull 

end 

say "Bye, bye..." 

exit 

Manchmal muss man eine Problemstellung auch mal von einer anderen Seite betrachten. Eine noch 
so einfache Entscheidung im Programm braucht Zeit. Wir erzeugen im Programmvorlauf eine Ent-
scheidungstabelle, in der alle möglichen Spielsituationen und Ergebnisse enthalten sind. Das 
braucht auch Zeit, aber nur einmal. Unterstellt wir, dass das Spiel mehrere Male nach Spielstart 
gespielt werden soll, wird die Tabelle nur einmal erzeugt und kann in jedem Durchgang das Ergebnis 
unmittelbar aus der Tabelle entnehmen. 

/* REXX * KNOBELN  

*/ 

obj.1="Stein" 

obj.2="Schere" 

obj.3="Papier" 

obj.4="Brunnen" 

obj.5="ENDE" 

obj.6="" 

obj.7="Triff Deine Auswahl (E für Ende)" 

erg.   ="Du hast gewonnen" 

erg.1.1="unentschieden" 

erg.2.2="unentschieden" 

erg.3.3="unentschieden" 

erg.4.4="unentschieden" 

erg.1.2="ich habe gewonnen" 

erg.2.3="ich habe gewonnen" 

erg.3.1="ich habe gewonnen" 

erg.3.4="ich habe gewonnen" 

erg.4.1="ich habe gewonnen" 

erg.4.2="ich habe gewonnen" 

do forever 

   say center("< Heinzi's kleine Knobelei >",79,'-') 

   do cnt = 0 to 7 

      say center(obj.cnt,79) 

   end 

   do until s >= 1 & s <= 4  

      parse external s 

      if translate(s) = "E" then exit 

   end 

   r=random(1,4) 

   say;say 

   say "Du hast" obj.s", ich habe" obj.r"," erg.r.s 

end 

exit 

1.29.4. Prüfziffernberechnung 

Bevor wir uns mit dem nächsten Thema auseinandersetzen, einige erklärende Worte zur Prüfziffern-
berechnung nach dem MODULO-11-Verfahren: Was ist das? Ein Rechenverfahren, das Tippfehler 
und Zahlendreher in numerischen Werten erkennbar machen soll. Man trifft es in der Praxis sehr 
häufig an, weiß es nur nicht. So nutzen es beispielsweise viele Banken, um ihre Kontonummern 
"sicherer" zu machen. Wenn ihr euch den Zahlenbandwurm einer ISBN (International Serial Book 
Number) anseht, ist die letzte Zahl daraus auch eine Prüfziffer nach dem MODULO-11-Verfahren. 
Könnte die ISBN in sich nicht auf Plausibilität geprüft werden, würde wohl kaum eine Buchbestellung 
über diese Nummer zu einem vernünftigen Ergebnis führen. Auch die Banken haben das Problem, 
dass ein simpler Tippfehler in einer Kontonummer möglichst nicht zu einer Falschbuchung führen 
sollte. Im Zweifelsfall ist es besser, den falschen Wert nicht zu verarbeiten, als eine Falschverarbei-
tung mit entsprechenden Stornos und Korrekturen hinnehmen zu müssen. 

Die Rechenformel am Beispiel der vierstelligen Zahl: 4812 

Die hinterste Stelle wird mit 2 multipliziert (4), die vorletzte Stelle mit 3 (3), die drittletzte Stelle mit 4 
(32), die viertletzte Stelle, in unserem Fall die erste, mit 5 (20). Anders ausgedrückt: Bei einer belie-
big langen Zahl wird die hinterste Stelle mit 2 multipliziert. Für jede Stelle, die es in der Zahl nach 
links geht, erhöht sich der Multiplikator um eins. Die ganzen Multiplikationsergebnisse werden ad-
diert. In unserem Fall ergäbe dies (4+3+32+20) 59. Diese Summe wird integer durch 11 dividiert. 
Ergebnis wäre 5 Rest 4. Wichtig ist nun, dass wir mit dem Rest weiterarbeiten müssen. Dieser Rest 
wird von 11 abgezogen, das Ergebnis ist die Prüfziffer. Theoretisch könnte diese Prüfziffer nun aber 
auch 10 oder 11 sein (bei Rest 0 oder Rest 1). Für den praktischen Einsatz darf die Prüfziffer aber 
nur ein Byte groß sein. Hier sieht eine Ausnahmeregel vor: Ist der ermittelte Wert größer als 9, wird 
als Prüfziffer 0 angenommen. 

Stellen wir uns nun folgende Situation vor: Wir arbeiten als Servicebetrieb für unterschiedliche Un-
ternehmen. Jedes dieser Unternehmen liefert täglich Daten zur Verarbeitung unter einer von uns 
vergebenen Firmennummer an. Die von uns vergebenen Firmennummern sind vierstellig und wir 
berechnen für diese Firmennummer eine Prüfziffer nach MODULO 11. Die Prüfziffer ergänzt die von 
uns vergebene Firmennummer, den gebildeten fünfstelligen Wert geben wir unserem Kunden als 
Kundennummer bekannt. 

Kundennummer=4812, Prüfziffer wird: 

((2 * 2)+(1 * 3)+(8 * 4)+(4 * 5)):11 = 5 REST 4 

11 - rest = 7 

Die ermittelte Prüfziffer ist 7, die Kundennummer lautet 48127. Diese teilen wir dem Kunden mit. 
Angenommen er liefert seine Buchungsdaten nun unter der Nummer 41827 (Zahlendreher), würden 
wir unter Umständen ohne Prüfverfahren die gelieferten Daten unter einem anderen Unternehmen 
verbuchen. Stornos, Rückbuchungen etc. sind mit entsprechendem Zeitaufwand unumgänglich. 
Besser wäre, wir würden vor Verarbeitung erkennen, dass die Firmennummer nicht stimmen kann. 
Wir gehen also her und führen eine Rückrechnung der Firmennummer durch. Da wir wissen, dass 
nur die ersten vier Stellen die wirkliche Firmennummer ergeben, prüfen wir: 

((2 * 2)+(8 * 3)+(1 * 4)+(4 * 5)):11 = 4 REST 8 

11 - REST = 3 

3 wäre die rückgerechnete Prüfziffer. Die Kundennummer müsste dementsprechend lauten: 41823, 
geliefert wurde aber unter 41827. Die gelieferte Kundennummer ist falsch. Wir können zwar nicht 
erkennen, wie die Kundennummer richtigerweise heißen müsste, werden aber die Daten, die unter 
falscher Kundennummer geliefert wurden, nicht verarbeiten und die Situation klären. Damit haben 
wir eine Fehlbuchung und ein entsprechend aufwendiges Korrekturverfahren umgangen. 

Sehen wir uns jetzt ein Programm an, das uns für eine beliebig lange Zahl die Prüfziffer nach MO-
DULO-11-Verfahren berechnet: 

/* REXX * PZ 

*/ 

do forever 

   do until datatype(zahl) = "NUM" ! length(zahl) = 0 

      say 'Zahl eingeben (ENTER = Ende)' 

      pull zahl 

   end 

   if length(zahl) = 0 then leave 

   zahl = abs(trunc(zahl)) 

   summe=0 

   pos=length(zahl) 

   multi=2 

   do length(zahl) 

      stelle_x = substr(zahl,pos,1) 

      mult_erg = stelle_x * multi 

      summe = summe + mult_erg 

      mult = mult + 1 

      pos = pos - 1 

   end 

   rest = summe // 11 

   pz = 11 - rest 

   if pz > 9 then pz = 0 

   say "Ergebnis mit Prüfziffer:" zahl"-"pz 

end 

say "Bye, bye..." 

exit 

Vereinfachen wir den Ablauf. Einige Befehle können zusammengezogen werden. Eine andere 
Schleifenvariante kommt uns zur Problemlösung weiter entgegen. Gehen wir davon aus, dass der 
Bediener das Programm aufruft und die zu berechnende Zahl bei Programmaufruf mitliefern soll. 
Das Programm spricht für sich selbst: 

/* REXX * PZ */ 

arg zahl . 

if datatype(zahl) = "CHAR" then do 

   parse source . . pgm_name . 

   say "Eingabefehler. Aufruf:" pgm_name "zahl" 

   exit 

end 

zahl = abs(trunc(zahl)) 

sum = 0 

do pos = 1 to length(zahl) 

   sum=substr(zahl,pos,1)*(length(zahl)+2-pos)+sum 

end 

pz=11-sum//11 

if pz > 9 then pz=0 

say zahl'-'pz 

EXIT 

1.29.5. Lottoziehung 

Eine schöne Übung für Variablenhandling ist ein Lottoprogramm. Viele haben es schon versucht, 
die meisten davon gehen immer noch mehr oder weniger mürrisch ihrer Arbeit nach. Solltet ihr mit 
dem Programm einen Haupttreffer landen, vergesst mich nicht... 

Sehen wir uns die Anforderungen an unser Programm an. Wir ziehen flexibel aus 45 oder 49 Zahlen. 
Die Anzahl zu ziehender Zahlen sollte variabel zwischen 6 (müssen es mindestens sein) und 12 (der 
größtmögliche Tipp im Vollsystem) sein. Die Ziehungszahlen werden über den Zufallsgenerator er-
mittelt. Ab der zweiten ermittelten Zahl müssen Prüfungen mit allen davor gezogenen Zahlen durch-
geführt werden. Wird erkannt, dass eine Ziehung die gleiche Zahl wie eine vorausgegangene Zie-
hung geliefert hat, muss die Ziehung wiederholt werden. 

Der Aufruf des Programmes könnte lauten: 

%LOTTO 9 aus 45 

In diesem Fall sollten 9 Zahlen zwischen 1 und 45 ermittelt werden. Werden keine oder falsche 
Programmparameter übergeben, wird für die Anzahl zu ziehender Zahlen 6, für den maximalen Wert 
(zwischen 1 und ?) 49 angenommen. 

/* REXX * LOTTO1  

*/ 

arg anz temp aus . 

if temp='TRACE' then trace '?r' 

if anz < 7 ! anz > 12 then anz = 6 

if aus >< 45 & aus >< 49 then aus = 49 

var.=0 

x=1 

do until x > anz 

  var.x=random(1,aus) 

   fehl=0 

   if x > 1 then do y=1 to x-1  /* Prüfen doppelte */ 

      if var.y = var.x then do 

         fehl=1 

         leave 

      end 

   end 

   if fehl=0 then x=x+1 

end 

erg_sort="" 

do x=1 to aus  /* Sortierte Zahlenausgabe */ 

   do y=1 to anz 

      if x=var.y then erg_sort = erg_sort var.y 

   end 

end 

say center("Heinzi's Lotto" anz 'aus' aus,79,'-') 

say;say;say 

say center(erg_sort,79) 

say;say;say 

say center('Viel Erfolg',79,'-') 

exit 

Relativ aufwändig. Letztlich eine nette Übung, mit Schleifen und Variablenstämmen in Form von 
Tabellen zu arbeiten. Versuchen wir uns aber zu vergegenwärtigen, wie die Ziehung der Lottozahlen 
im wirklichen Leben abläuft: Die Trommel wird mit der Anzahl Kugeln gefüllt. Der Lottogott bestimmt 
innerhalb der Trommel die Position der Kugel, die gezogen werden soll. Die entsprechende Kugel 

wird entnommen. Die Anzahl der Kugeln in der Trommel verringert sich. Die Kugel, welche die Trom-
mel verlassen hat, kann kein zweites Mal gezogen werden. So einfach ist das. Machen wir es also 
auch so! Wir nutzen eine Wortfolge innerhalb eines Strings wie eine Tabelle: 

/* REXX * LOTTO2 */ 

arg anz temp aus . 

if left(temp,1)="T" then trace "?r" 

if anz < 7 ! anz > 12 then anz = 6 

if aus >< 45 & aus >< 49 then aus = 49 

trommel="" 

kugeln="" 

do idx=1 to aus 

   trommel=trommel" "idx 

end 

do idx=1 to anz 

   rnd=random(1,words(trommel)) 

   kugeln=kugeln word(trommel,rnd) 

   trommel=delword(trommel,rnd,1) 

end 

erg_sort="" 

do x=1 to aus  /* Sortierte Zahlenausgabe */ 

   do y=1 to words(kugeln) 

      if x=word(kugeln,y) then 

      erg_sort=erg_sort word(kugeln,y) 

   end 

end 

say center(erg_sort,79) 

say 

say center('Viel Erfolg',79,'-') 

exit 

1.29.6. Mastermind 

Nochmal ein Spiel (irgendwann werden wir Dinge versuchen, die erschreckende Ähnlichkeit mit Ar-
beit haben). Interessante Tabellenarbeit kommt in Mastermind vor. 

Zum Spiel: Vier farbige Spielknöpfe werden von einem Spieler verdeckt gelegt. Der zweite Spieler 
versucht, die Farbkombination zu erraten, indem er ebenfalls vier farbige Spielsteine auflegt. Spieler 
1 teilt nach jedem Versuch von Spieler 2 mit, wie viele Spielsteine mit richtiger Farbe an der richtigen 
Position und wie viele Spielsteine mit richtiger Farbe an falscher Position vorkommen. Wir können 
unter TSO/REXX leider nicht mit farbigen Steinen spielen (mit Dialog Manager kein Problem, sofern 
Farbterminals vorhanden sind), können das ganze aber mit Zahlen nachvollziehen. Der Rechner 
ermittelt eine vierstellige Zahl, der Spieler muss sie erraten. Ist das Programm flexibel ausgelegt, 
spielt es keine Rolle, ob wir mit drei-, vier- oder fünfstelligen Zahlen spielen. Das Programm kommt 
mit jeder Zahl gleichermaßen zurecht. 

/* REXX * MASTER  

Spielt Mastermind gegen einen Spieler 

Der Spieler hat 10 Lösungsversuche. 

Bei jedem Versuch wird mitgeteilt: 

Richtige Zahlen am richtigen Ort 

Richtige Zahlen am falschen Ort 

*/ 

arg max test . 

if test = "T" then trace "?R" 

if datatype(max) = "CHAR" then max = 9999 

do forever 

   v=0 

   zahl0=right(random(0,max),length(max),0) 

   do until v >= 10 ! zahl2 = zahl0 

      ro=0 

      rz=0 

      zahl1=zahl0 

      say v+1'. Eingabe:' 

      pull zahl2 

      do while datatype(zahl2) = char ! zahl2 > max 

         say 'Eingabe falsch, wiederholen' 

         pull zahl2 

      end 

      zahl2=right(zahl2,length(max),0) 

      v=v+1 

      if zahl1=zahl2 then do /* Lösung richtig */ 

         say "Gewonnen im" v". Versuch" 

      end 

      else do             /* Lösung falsch  */ 

         do x=1 to length(zahl1) 

            if substr(zahl2,x,1)=substr(zahl1,x,1)  

            then do 

               ro=ro+1 

               zahl1=overlay('a',zahl1,x,1) 

               zahl2=overlay('b',zahl2,x,1) 

            end 

         end 

         do x=1 to length(zahl1) 

            do y=1 to length(zahl1) 

               if substr(zahl1,x,1)=substr(zahl2,y,1) 

                 then do 

                 zahl1=overlay('a',zahl1,x,1) 

                 zahl2=overlay('b',zahl2,y,1) 

                 rz=rz+1 

              end 

           end 

        end 

        if v < 10 then say 'Im' v'.Versuch:' , 

           'Zahlen am richtigen Ort =>' ro, 

           'Zahlen am falschen Ort =>' rz 

         else say 'erst denken dann eingeben' 

      end 

   end 

   do until ant = 'J' ! ant = 'N' 

      say 'Ein neues Spiel (J/N):' 

      pull ant 

   end 

   if ant = 'N' then leave 

end 

say "Bye, bye..." 

exit 

1.29.7. CD-Wechsler 



Inside z/OS Seite 107 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

Ein ganz anderes Problem. Viele CD-Wechsler wählen über einen Randomizer zufällig Musikstücke. 
Da wir ja wissen, dass es in der EDV keine Zufälle gibt, lässt sich natürlich auch ein Zufallsgenerator 
so steuern, wie wir es für unsere Bedürfnisse brauchen (nicht 2mal in 30 Minuten das gleiche Stück 
aus 150). Von einem vernünftigen Zufallsgenerator an dieser Stelle würde ich erwarten, dass er 
zufällig einen Titel wählt und diesen erst dann wieder berücksichtigt, wenn er alle anderen verfüg-
baren Titel einmal ausgewählt hat. Erst wenn jeder Titel einmal gespielt wurde, darf der Zufallsge-
nerator wieder aus allen verfügbaren Stücken wählen. Legen wir zugrunde, dass über einen Fest-
wertspeicher die aktuelle Umgebung der Stücke-Auswahl hinterlegt werden kann. Wenn das Radio 
ausgeschaltet wird, kann bei erneutem Einschalten wieder dort weitermusiziert werden, wo unter-
brochen wurde. Insgesamt möchte ich natürlich der Firma SONY meinen Dank aussprechen, für 
das, was sie im Bereich Unterhaltungselektronik so leistet, aber auch anregen, etwas mehr Intelli-
genz in die CD-Ansteuerung zu implementieren. So könnte es in groben Zügen aussehen: 

/* REXX * CD  

*/ 

say "Zufallsgenerator für CD-Steuerung" 

do until nochmal = "N" 

   max=6 

   cd.1="DE_BURGH Spaceman Getaway Emotion Borde" 

   cd.2="DIAMOND Caroline Shilo Play_me Stones Holly" 

   cd.3="FENDRICH Cyrano Papa TuttiFrutti Macho" 

   cd.4="MAY Wolken Klempner Annabelle Buffet Hilf " 

   cd.5="HAINDLING Depp Baby Hellseher Flipper Samba" 

   cd.6="FENDRICH Wind Regen Nix_fix Westen Scene" 

   do while max > 0 

      nr=random(1,max) 

      cd=cd.nr 

      ti=random(2,words(cd.nr)) 

      say "Gewählt:" word(cd.nr,ti) , 

          "von" word(cd.nr,ti) 

      cd.nr=delword(cd.nr,ti,1) 

      if words(cd.nr) = 1 then do 

         cd.nr=cd.max 

         max=max - 1 

      end 

      say "Wiederholung (J/N)" 

      parse upper external nochmal 

end 

say "Bye, bye..." 

exit 

1.29.8. Just_4_Fun 

Immer wieder lustig ist eine Situation, die man heute sehr nobel einen Know-how-Transfer nennt. 
Man sieht sich in den Programmbibliotheken der Kollegen um und sucht dort nach interessant er-
scheinenden kleinen Helfern. Wird man fündig, wird kopiert und der Namen des Programmautors 
entsprechend verändert. Trifft man dabei auf übersetzte REXX-Programme, lässt sich leider nicht 
mehr vorab erkennen, was ein Programm macht. Macht nichts, man kann sich in vielen Fällen an 
den Programmnamen orientieren. Trifft man auf Programmnamen wie GEHEIM oder VORSICHT, 
üben diese eine schier unwiderstehliche Anziehungskraft aus. Nun ist Adrenalin ein wichtiges Hor-
mon für den Körper, und wir werden versuchen, unsere lieben Kollegen damit ein bisschen zu ver-
sorgen: 

/* REXX * GEHEIM  

*/ 

say "Achtung, Sie starten FREE-SPACE, das neue" 

say "Dienstprogramm der CRG (Computer Revolution Group)" 

say "Platzoptimierung ist kein Vorrecht des PC's" 

say 

say "Programm wirklich starten?" 

do until ant = "J" ! ant = "N" 

   pull ant 

end 

if ant = "N" then exit 

dmy=OUTTRAP("file.") 

"LISTC" 

do x=2 to file.trapped 

   say "ENTRY (A)" file.x "DELETED" 

end 

say "Danke für den Einsatz von FREE-SPACE-Utility" 

say "Bye, bye......" 

EXIT 

1.29.9. Dynamic Allocation (1) 

In den nächsten drei Beispielen wollen wir uns, wie versprochen, mit der dynamischen Erweiterung 
einer bestehenden Zuweisung auseinandersetzen. Das erste Beispiel ergänzt die Zuweisung für 
SYSEXEC (REXX-Programmaufrufe). Standardmäßig wird die bestehende Zuweisung um die Datei 
userid.MVS.EXEC erweitert. Die Datei wird in der Suchreihenfolge von SYSEXEC vorne eingescho-
ben (könnte Probleme geben, wenn die verketteten Dateien unterschiedliche Blockung aufweisen). 

/* REXX ** ADD userid.P.EXEC to SYSEXEC */ 

call outtrap "var." 

"lista st" 

call outtrap "OFF" 

do §=1 to var.0 

   if pos("SYSEXEC",substr(var.§,3,8)) > 0 then leave 

end 

if § <= VAR.0 then do 

   ds_lst = "'"value("VAR."§-1)"'" 

   do _1=§+1 to var.0 by 2 

      _2=_1+1 

      if substr(var._2,3,8) >< " " then leave 

      ds_lst = ds_lst ",'"var._1"'" 

   end 

end 

ds_lst = space("'"userid()".P.EXEC',"ds_lst,0) 

"executil searchdd(no) execdd(close)" 

"alloc dd(SYSEXEC) da("ds_lst") shr reuse" 

"executil searchdd(yes)" 

Exit 

1.29.10. Dynamic Allocation (2) 

Das nächste Beispiel lassen wir für sich selbst sprechen. Im Grunde haben wir das gleiche Kern-
problem bei dynamischer Ergänzung der Dialog Manager Zuweisungen. Wird an dieser Stelle 
schlecht gearbeitet, gibt es immer lange Gesichter. Versauen wir unsere SYSEXEC-Zuweisung, sind 

schlimmstenfalls einige Programmaufrufe nicht mehr möglich. Verschlimmbessern wir das DMS-
Umfeld, können wir nicht mehr mit PDF arbeiten. Für dynamische Veränderungen im Dialog Mana-
ger müssen wir sechs DD-Statements berücksichtigen. Vier davon müssen erweitert, zwei müssen 
gegebenenfalls umbelegt werden (Die Zuweisungen ISPTABL und ISPFILE dürfen immer nur auf 
eine Datei zeigen). Diese Routine läuft automatisch bei LOGON ab, wenn ein Service, wie er in den 
meisten großen Unternehmen von der Systemprogrammierung angeboten wird, genutzt werden 
kann: In der standardmäßig zugewiesenen SYSPROC-Datei userid.???.CLIST wird ein Member 
START oder namensgleich mit der Benutzerkennung hinterlegt. Dieses Member wird bei LOGON 
gesucht. Wird es gefunden, werden die Befehle, die es enthält, ausgeführt. Nun sollten in einer 
CLIST-Datei auch CLISTen stehen. Die CLIST könnte aber ihrerseits diese REXX-Routine, wie auch 
das vorausgegangene Beispiel für SYSEXEC-Erweiterung, aufrufen. 

By the way: Dies sollte vor dem Start von ISPF erfolgen (in der Regel im Zuge von LOGON). Wäre 
ISPF bereits geladen, würde man diese Zuweisungen mittels LIBDEF-Service durchführen. 

/* REXX ** ISPALLOC 

   Erweiterte Zuweisungen für DMS 

   ISPPLIB ISPMLIB ISPSLIB und ISPTLIB werden ergänzt 

   ISPFILE und ISPTABL neu festgelegt 

 

   Funktioniert nur während des LOGONs solange noch keine ISPF- 

   Umgebung geladen ist. 

*/ 

if sysvar("SYSISPF") = "ACTIVE" then do 

   say "Programm nicht für die Ausführung unter ISPF gedacht" 

   exit 

end 

uid = userid() 

call outtrap "LISTA." 

"lista st" 

call outtrap "OFF" 

_=0 

_=_+1;alo._="ISPPLIB '"uid".MURKS.PANELS','"uid".TEST.PANELS'" 

_=_+1;alo._="ISPSLIB '"uid".TEST.SKELS'" 

_=_+1;alo._="ISPMLIB '"uid".TEST.MSGS'" 

_=_+1;alo._="ISPTLIB '"uid".TEST.TABLES'" 

_=_+1;alo._="ISPTABL '"uid".TEST.TABLES'" 

_=_+1;alo._="ISPFILE '"uid".TEST.SKELS'" 

ds_lst.= 

do _1=1 to _ 

   parse var alo._1 dd dsl 

   do §=1 to lista.0 

      First10 = word(substr(lista.§,1,10),1) 

      Select 

         when pos("NULLFILE",First10) > 0 then iterate 

         when pos("DDNAME"  ,First10) > 0 then iterate 

         when wordpos(dd,First10)     > 0 then leave 

         otherwise iterate 

      end 

   end 

   if § < lista.0 then do 

      ds_lst.dd = "'"value("lista."§-1)"'" 

      do _2=§+1 to lista.0 by 2 

         _3=_2+1 

         if substr(lista._3,3,8) >< " " then leave 

         ds_lst.dd = ds_lst.dd",'"lista._2"'" 

      end 

   end 

   if wordpos(dd,"ISPTABL ISPFILE") > 0 then AloDs = dsl 

   else                   AloDs = ds_lst.dd","dsl 

    "alloc dd("dd") da("AloDs") shr reuse" 

End 

Exit 

1.29.11. Dateiverarbeitung 

In den nächsten Beispielen werden wir uns mit Dateiverarbeitung auseinandersetzen. Wir werden 
uns einfaches Lesen, Lesen aus mehreren Dateien, Lesen aus und Schreiben in mehrere Dateien 
in einem Programm, und den Update (Lesen und Schreiben in die gleiche Datei in einem Arbeits-
gang) ansehen.  

1.29.11.1. Anlisten einer Datei (einfaches Lesen) 

Zunächst eine ganz kleine Demonstration für Dateiausgabe auf Bildschirm (vergleichbar mit TYPE 
datei unter DOS - Verzeihung). 

/* REXX  

Zeigt Inhalt einer Datei am Bildschirm 

*/ 

arg datei . 

if length(datei) = 0 then call fehler 

dmy=msg(off) 

"alloc dd(list) da("datei") shr reuse" 

if rc > 0 then call fehler 

"execio * diskr list (finis" 

do queued() 

   parse pull satz 

   say satz 

end 

"free dd(list)" 

say "Bye, bye..." 

exit 

fehler: 

say "Datei nicht angegeben, oder Zuweisungsfehler" 

exit 

1.29.11.2. Datei-Inhalt an TSO-User senden 

Nutzen wir das Verfahren, um die vorbereiteten Inhalte eines bestimmten PO-Members an beliebige 
TSO-Benutzer als Nachrichten zu versenden. 

/* REXX * SEND  

Liest den Inhalt eines PO-Members übermittelt die Sätze  

Zeilen weise an einen oder mehrere TSO-User 

*/ 

arg member empf 

dummy=msg(off) 

if length(empf)=0 then signal fehler 

if length(member)=0 then member='STDMSG' 

"alloc dd(msg) da(msgs.data("member")) shr reuse" 

if rc > 0 then signal fehler 

"execio * diskr msg (finis" 

"free dd(msg)" 

do queued() 

   parse pull msg 

   msg=left(msg,71) 

   "se '"msg"' user("empf") logon" 

end 

say;say "SEND an" empf "ausgeführt.." 

exit 

 

fehler: 

   "free dd(msg)" 

   say "Zuweisungsfehler, oder Empfänger fehlt 

exit 

1.29.11.3. Lesen, Output Stack 

Lesen aus einer Datei in den Stackbereich mit wiederholter Verarbeitung. 

/* REXX * LIST  

Suchen wahlfreier Begriffe in einer beliebigen Datei 

*/ 

arg trace . 

if trace="T" then trace '?r' 

else call msg "OFF" 

say 'Datei vollqualifiziert (ENTER = Ende)' 

do forever 

   parse upper external datei 

   if datei='' then exit 

   if sysdsn("'"datei"'") >< OK then 

      msg='Datei nicht vorhanden' 

   else do 

      call listdsi "'"datei"'" 

      select 

         when sysdsorg="PO" & index(datei,'(')=0 then 

            msg='PO-Format, Membername fehlt' 

         when wordpos(sysdsorg,"PO PS")= 0 then 

            msg='Format' sysdsorg 'nicht möglich' 

         otherwise leave 

      end 

   end 

   say msg ', bitte nochmal' 

end 

"qstack" 

if queued() > 0 ! rc > 1 then "newstack" 

"alloc dd(such) da('"datei"') shr reuse" 

"execio * diskr such (finis" 

"free dd(such)" 

do forever 

   say 'Suchbegriff angeben:' 

   parse upper external such 

   if length(such) = 0 then leave 

   c_found='0' 

   msg='Suche nach' such 'in' datei 

   say center(msg,79,'*') 

   say 

   do c_satz = 1 to queued() 

      parse pull satz 

      if pos(such,translate(satz)) > 0 then do 

         say "("right(c_satz,3,"0")")" satz 

         c_found=c_found+1 

      end 

      queue satz 

   end 

   say 

   if c_found=0 then msg=such 'nicht gefunden' 

   else msg='Listende, gefundene Sätze:' c_found 

   say center(msg,79,'*') 

end 

"delstack" 

say "Bye, bye..." 

exit 

1.29.11.4. Lesen, Output Stem 

Lesen aus einer Datei in einen Variablenstamm. Um nicht immer das Gleiche zu sehen, hier eine 
Variante, die uns erlaubt, bei der Anzeige größerer Dateien rückwärts zu blättern. Dies ist nur mög-
lich, weil die Datei im Programm als Variablenstamm vorliegt, innerhalb dessen beliebig positioniert 
werden kann. EXECIO selbst erlaubt nur eine starre, sequentielle Verarbeitung. Das heißt: lesen wir 
den 7. Satz und wollen zurück zum 5., muss die Datei geschlossen, erneut geöffnet und vorgelesen 
werden. 

/* REXX * SHOW  

Zeigt den Inhalt einer Datei seitenweise am 

Bildschirm  und erlaubt dem Anwender, 

in der Datei vorwärts  oder rückwärts zu blättern 

Bedingung: CLRSCRN-Befehl funktioniert (Clear Screen) 

*/ 

arg datei test . 

if test = "T" then trace "?R" 

if length(datei) = 0 then do 

   say "Dateiname muss bei Aufruf angegeben werden" 

   exit 

end 

do forever 

   if sysdsn(datei) >< 'OK' then 

      msg = 'Datei' datei 'nicht verfügbar' 

   else do 

      x=listdsi(datei) 

      select 

         when sysdsorg='PO' & pos("(",datei)=0 then 

            msg='PO-Datei ohne Member' 

         when wordpos(sysdsorg,"PO PS")=0 then 

            msg='DSORG wird nicht unterstützt' 

         otherwise leave 

      end 

   end 

   say msg 

   parse upper external datei 

end 

"alloc dd(list) da("datei") shr reuse" 

"execio * diskr list (stem satz. finis" 

"free dd(list)" 

x=1 

do forever 

   clrscrn /* nicht in allen Unternehmen verfügbar */ 

   say center('< 'datei' >',79,'-');say 

   do y=x to x+17 

      say right(y,4,0)'..'left(satz.y,73,) 

      if y >= satz.0 then leave 

   end 

   say;say 'F<oreward>   B<ack>   E<nd>' 

   do until ant = 'F' ! ant = 'B' ! ant = 'E' 

      pull ant 

   end 

   select 

      when ant='F' then x=y-1 

      when ant='B' then do 

         x=y-35 

         if x < 1 then x=1 

      end 

      otherwise leave 

   end 

end 

say "Bye, bye..." 

exit 

1.29.11.5. Mischen (Merge) 

Bisher alles Kinderkram, machen wir mal was Vernünftiges. Legt Papier und Schere bereit. Wir bas-
teln uns ein Mischprogramm. Zwei Datenbestände zu einem dritten zusammenzumischen erfordert 
bestimmte Voraussetzungen: Beide Dateien müssen gleich strukturiert und nach dem Mischkrite-
rium aufsteigend sortiert vorliegen. Die beiden Eingaben werden parallel zueinander verarbeitet. Je 
ein Satz aus beiden Dateien wird verglichen. Der kleinere Datensatz (der Satz, dessen Mischkrite-
rium das kleinere ist) wird in die Ergebnisdatei geschrieben. Aus der Datei, deren Satz geschrieben 
wurde, wird der nächste Datensatz gelesen. Ist eine der beiden Dateien zu Ende, werden die Rests-
ätze der anderen Datei noch in die Ergebnisdatei überführt. Das Programm sollte möglichst so an-
gesetzt werden, dass es eine Gleichbehandlung aller Datensätze bis zum Schluss erlaubt, gleich-
gültig, welche der beiden Eingabedateien zuerst auf Dateiende läuft (alles andere hat nichts mit 
Mischen zu tun, sondern wäre besser in die Rubrik "Schütteln" einzureihen). 

Gehen wir nun für unser Beispiel von einigen Dingen aus: Die beiden Eingaben sind die Member 
MISCH1 und MISCH2 der Datei SP1.EDU.DATA. Das Ergebnis wird im Member MISCHERG der 
gleichen Datei hinterlegt. Die Eingaben liegen nach Artikelnummern sortiert aufsteigend vor (Stelle 
1 bis 4 des Datensatzes). Die Artikelnummer ist das Mischkriterium. 

Die folgende Lösung wurde nach rein logischen Gesichtspunkten kodiert und wäre in dieser Form 
nahezu unverändert in jede andere Programmiersprache übertragbar. Sorry..., aber wie man es mit 
allen Vorzügen von REXX anders machen kann, zeigt dann das erweiterte zweite Beispiel. 

/* REXX * MISCH  

Mischt zwei Eingaben zu einer neuen Ausgabe zusammen.  

Mischkriterium ist eine Artikelnummer in Stelle 1 bis 4 der Eingaben. 

*/ 

parse upper arg trace . 

if trace="T" then trace '?r' 

else call msg "OFF" 

"free dd(m1 m2 m3)" 

"alloc dd(m1) da(kurs.data132(misch1)) shr reuse" 

"alloc dd(m2) da(kurs.data132(misch2)) shr reuse" 

"alloc dd(m3) da(kurs.data132(mische)) old" 

s_eof_m1 = 0 

s_eof_m2 = 0 

s_les_m1 = 1 

s_les_m2 = 1 

do forever 

   if s_eof_m1 = 0 then do 

      "execio 1 diskr m1" 

      if rc = 2 then do 

         s_eof_m1 = 1 

         art_m1 = "ff"x 

      end 

      else do 

         parse pull satz_m1 

         art_m1=left(satz_m1,4) 

         s_les_m1 = 0 

      end 

   end 

   if s_eof_m2 = 0 then do 

      "execio 1 diskr m2" 

      if rc = 2 then do 

         s_eof_m2 = 1 

         art_m2 = "ff"x 

      end 

      else do 

         parse pull satz_m2 

         art_m1=left(satz_m2,4) 

         s_les_m1 = 0 

      end 

   end 

   if s_eof_m1!!s_eof_m2 = 11 then leave 

   select 

      when art_m1 < art_m2 then do 

         queue satz_m1 

         s_les_m1 = 1 

      end 

      when art_m1 > art_m2 then do 

         queue satz_m2 

         s_les_m2 = 1 

      end 

      otherwise do 

         queue satz_m1 

         queue satz_m2 

         s_les_m1 = 1 

         s_les_m2 = 1 

      end 

   end 

   "execio" queued() "diskw m3" 

end 

"execio 0 diskr m1 (finis" 

"execio 0 diskr m2 (finis" 

"execio 0 diskw m3 (finis" 

"free dd(m1 m2 m3)" 

exit 

Jetzt legen wir noch eins drauf: Werden gleiche Artikelnummern in beiden Dateien erkannt, darf nur 
der Datensatz aus MISCH1 in die Ergebnisdatei geschrieben werden. Der Satz aus MISCH2 muss 
übersprungen werden. Ein Fehlerprotokoll muss erzeugt und im Member MISCHF der gleichen Datei 
hinterlegt werden. Traten keine gleichen Artikelnummern auf, sollte auch dies in der Protokolldatei 
erkennbar sein. 

Gleichzeitig ändern wir unser Programm ein wenig ab und nutzen bei der Kodierung die Vorzüge 
von REXX, was den Ablauf wesentlich vereinfacht. Die beiden Eingabedateien hinterlegen wir in 
Variablenstämmen, das Protokoll bereiten wir in einem Variablenstamm vor, die Ergebnisdatei wird 
im Stack zum Schreiben vorbereitet. 

/* REXX * MISCH  

Mischt zwei Eingaben zu einer neuen Ausgabe zusammen. 

Mischkriterium ist eine Artikelnr in Stelle 1 bis 4 

Gleiche Artikelnummern werden protokolliert 

*/ 

parse upper arg trace . 

if trace="T" then trace '?r' 

else call msg "OFF" 

"free fi(m1 m2 m3 m4)" 

"alloc fi(m1) da(edu.data(misch1)) shr reuse" 

"alloc fi(m2) da(edu.data(misch2)) shr reuse" 

"alloc fi(m3) da(edu.data(mische)) old" 

"alloc dd(m4) da(edu.data(mischf)) old" 

"execio * diskr m1 (stem m1. finis)" 

"execio * diskr m2 (stem m2. finis)" 

interpret 'm1.'m1.0+1' = "ff"x' 

interpret 'm2.'m2.0+1' = "ff"x' 

x=1 

y=1 

z=0 

do forever 

   select 

      when left(m1.x,4) < left(m2,y,4) then do 

         queue m1.x 

         x=x+1 

      end 

      when left(m1.x,4) > left(m2,y,4) then do 

         queue m2.y 

         y=y+1 

      end 

      otherwise do 

         if m1.x = "ff"x then leave 

         queue m1.x 

         z=z+1 

         m3.z="Satz doppelt" , 

              "(M1:" x "M2:" y "M2-Satz folgt" 

         z=z+1 



Inside z/OS Seite 116 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

         m3.z=m2.y 

         x=x+1 

         y=y+1 

      end 

   end 

end 

queue 

if z = 0 then m3.1 = "Keine doppelten Sätze erkannt" 

"execio * diskw m3 (stem m3. finis" 

"execio * diskw m4 (finis" 

exit 

1.29.11.6. Datei-Update 

Wenden wir uns dem letzten zu, was uns die Dateiverarbeitung anzubieten hat: dem UPDATE. Unter 
UPDATE verstehen wir nicht, eine Datei komplett in den Stack zu lesen, dort einzelne Sätze zu 
verändern und anschließend den gesamten Stack wieder zu schreiben und dabei die alte Datei zu 
überschreiben. Vielmehr geht es darum, eine Datei satzweise zu lesen, zu erkennen, ob der gele-
sene Satz verändert werden soll, die Änderung durchzuführen und den Satz in seiner geänderten 
Form dorthin in die Datei zurückzubringen, von wo er gelesen wurde. 

Das folgende Beispiel zeigt ein Programm, das in einer Artikeldatei nach bestimmten Sätzen sucht 
und für diese Sätze eine Preisaktualisierung durchführt. Wir unterstellen, dass die Datensätze in den 
ersten sechs Stellen eine Artikelnummer enthalten. Die Änderung könnte über eine gezielte Artikel-
nummer oder einen Teil davon (Eingabe 01 soll alle Sätze aktualisieren, bei denen die Artikelnum-
mer mit 01 beginnt) erfolgen. Der Preis steht rechtsbündig zwischen den Stellen 56 und 65. 

/* REXX * PREISUPD  

Rechnet einen prozentualen Auf- oder Abschlag für  

angegebene Artikelnummern. Gesucht wird nach  

vollständigen Artikelnummern oder Teilen  

(linksbündig) davon.  

Wird ein zu ändernder Satz gefunden, wird er  

geändert in die Datei zurückgeschieben 

*/ 

arg test . 

if test = 'T' then trace '?r' 

dmy=msg(off) 

"free dd(upd)" 

"alloc dd(upd) da(edu.data(preise)) old" 

say center('< Preis-Änderungsdienst >',79,'-') 

say;say;say 

say "Artikelnummer eingeben (mindestens 2 Stellen)" 

do until datatype(artnum) = 'NUM' & length(artnum) <= 2 

   say "(numerisch)" 

   parse external artnum 

end 

artnum=trunc(artnum) 

say;say 

say "Preisänderung in %" 

do until datatype(aend)='NUM' 

   say "(z.B.: 5.7 oder -3)" 

   parse external aend 

end 

ergebnis="Kein Satz zu Nummer" artnum "gefunden." 

satzupd=0 

"execio 1 diskru upd" 

do while RC = 0 

   parse pull satz 

   if left(satz,length(artnum)) = artnum then do 

      preisneu=substr(satz,56,10)*((100+aend)/100) 

      satz=left(satz,55)right(format(preisneu,,2),10) 

      queue satz 

      "execio 1 diskw upd" 

      satzupd=satzupd + 1 

   end 

   "execio 1 diskru upd" 

end 

"execio 0 diskw upd (finis" 

"free dd(upd)" 

if satzupd > 0 then 

   ergebnis="Durchgeführte Änderungen:" satzupd 

say ergebnis 

exit 

Das zweite Update-Beispiel soll undokumentiert auf euch wirken. Versucht selbst nachzuvollziehen, 
was in diesem Programm vor sich geht. 

/* REXX * TELUPD 

*/ 

"alloc dd(upd) da(edu.data(telefon2)) shr reuse" 

say center(" Änderung Telefonverzeichnis ",78,"*");say;say 

say "Suchbegriff eingeben" 

parse upper external such 

if such = "" then exit 

say "Neue Telefonnummer" 

parse external aend 

c_found=0 

c_change=0 

do forever 

   "execio 1 diskru upd" 

   if rc=2 then leave 

   parse pull telefon 

   if pos(such,left(translate(telefon),51)) = 0 then iterate 

   c_found=c_found+1 

   Vorwahl = word(telefon,words(telefon)-1) 

   Nummer  = word(telefon,words(telefon)) 

   VwPos = pos(Vorwahl,Telefon) 

   NrPos = pos(Nummer,Telefon) 

   if left(aend,1) = "(" then parse var aend Vorwahl Nummer 

   else                       Nummer = Aend 

   alt = strip(telefon,"T") 

   neu = left(alt,51)left(Vorwahl,NrPos-VwPos)strip(Nummer) 

   say "Alt =>" alt 

   say "Neu =>" neu 

   say "U=Update I=Ignorieren" 

   do until ant = "U" ! ant = "I" 

      parse upper external ant 

   end 

   if ant = "U" then do 

      queue neu 

      "execio 1 diskw upd" 

      c_change=c_change+1 

   end 

end 

"execio 0 diskw upd (finis" 

Say 

say center(" Gefunden:" c_found", Geändert:"c_change" ",78,"*") 

exit 

1.29.12. Glücksrad 

Das nächste Programm zeigt uns, was viele Menschen im heute biblischen Alter jenseits der 50 
damals im Kindesalter als Zeitvertreib kannten (es soll tatsächlich in grauer Vorzeit nicht in jedem 
Kinderzimmer ein PC gestanden haben). Heute lebt dieser Zeitvertreib in TV-Sendungen wie 
Glücksrad wieder auf: WORTE RATEN. Wir holen uns dazu die zu ratenden Worte aus einer Datei. 
Dies hat den Vorzug, dass die Wortauswahl beliebig erweitert werden kann, ohne am Programm 
rumfummeln zu müssen. Aus den gelesenen Worten wird jeweils eines per Zufallsgenerator ausge-
wählt (entspricht letztendlich dem wahlfreien Zugriff auf eine Datei). Ein ausgewähltes Wort soll in-
nerhalb des Spieles kein zweites Mal benutzt werden.  Das gesuchte Wort wird vorgelegt, wobei an 
Stelle der Buchstaben für den Spieler nur Punkte zu sehen sind. Der Anwender kann einen Buch-
staben eingeben, und es wird ihm gezeigt, ob und an welchen Positionen im Wort der Buchstabe 
vorkommt. Dass ein Buchstabe vom Anwender mehr als einmal benutzt wird, wird vom Programm 
ausgeschlossen. Welche Buchstaben noch zum Raten zur Verfügung stehen, wird gezeigt (jeder 
benutzte Buchstabe wird aus der Reihe der noch zur Verfügung stehenden Buchstaben ausge-
schlossen). Meint der Anwender, das Wort erraten zu haben, kann er das Lösungswort auch voll-
ständig eintippen. 

/* REXX *  

Der Rechner liest Worte aus einer  Datei und  

wählt sie über Zufallsgenerator aus. Der Spieler 

muss das Wort erraten.  

*/ 

arg test . 

if abbrev(test,'T') then trace '?r' 

"alloc fi(wort) da('sp1.edu.data(worte)') shr reu" 

"execio * diskr wort (stem wort. finis" 

do forever 

   x=1 

   abc=space('ABCDEFGHIJKLMNOPQRSTUVWXYZ',1) 

   if wort.0=0 then leave 

   i=random(1,wort.0) 

   wort1=space(translate(wort.i),0) 

   wort2=copies('.',length(wort1)) 

   do until wort2 = wort1 

      say center("< Heinzi's Worteraten >",79,'-') 

      say;say;say 

      say center(wort2,79) 

      say;say;say 

      say 'Geben Sie einen Buchstaben, oder das' , 

          'Lösungswort ein (*=Ende)' 

      say abc 

      say;say 

      say x'. Versuch';say 

      pull ant 

      select 

         when ant='*' then do 

            say "Das gesuchte Wort war" wort1 

            say "Schäm Dich..." 

            exit 

         end 

         when ant < 'A' ! ant > 'Z' then do 

            say 'Nur ein Buchstabe oder die Lösung' 

            say 'weiter mit <ENTER>' 

            pull 

            iterate 

         end 

         when x > 1 & length(ant) = 1 & , 

            pos(ant,abc) = 0 then do 

            say 'Buchstabe nicht mehr möglich' 

            say 'weiter mit <ENTER>' 

            pull 

            iterate 

         end 

         otherwise if length(ant) = 1 then call check 

      end 

      if ant=wort1 ! wort2=wort1 then do 

         say 'Gut!' 

         say 'Lösung im' x'. Versuch' 

         loesung='' 

         do x=1 to length(wort1) 

            loesung=loesung!!substr(wort1,x,1)' ' 

         end 

         say;say 'Lösung:' loesung;say;say 

         say 'weiter mit <ENTER>' 

         pull 

         leave 

      end 

      x=x+1 

   end /* until wort1 = wort2 */ 

   max=wort.0 

   wort.i=wort.max 

   wort.0=wort.0-1 

end 

say 'Bye, bye...' 

exit 

/* Beginn SUBROUTINE CHECK **********************/ 

check: 

do y=1 to length(wort1) 

if substr(wort1,y,1) = ant then 

wort2 = overlay(ant,wort2,y,1) 

end 

if index(abc,ant) > 0 then 

abc= overlay('-',abc,index(abc,ant),1) 

return 

/* Ende SUBROUTINE CHECK ************************/ 

1.29.13. Job Control Generierung 

Im Folgenden sehen wir uns ein paar Beispiele für Job-Control-Verarbeitung mit REXX an. Im ersten 
Beispiel werden die JCL-Sätze für einen Druckauftrag mit IEBGENER im Stack hinterlegt und zur 
Ausführung an den INTERNAL READER übergeben. 

/* REXX * JCLPRT1  

Hinterlegt JCL im Stack und führt SUBMIT-Befehl aus. 

*/ 

arg test . 

if abbrev(test,'T',1) then trace('?r') 

else dmy=msg("OFF") 

say 'Abrechnungsnummer (JOB-ACCOUNT) angeben:' 

parse upper external actnum 

do forever 

   say 'Druckdatei vollqualifiziert (ENTER = Ende)' 

   do forever 

      parse upper external datei 

      if length(datei) = 0 then exit 

      if sysdsn("'"datei"'")="OK" then do 

      dmy=listdsi("'"datei"'") 

      select 

         when sysdsorg >< "PS" & sysdsorg >< "PO" , 

            then msg = "Dateiorganisation falsch" 

         when sysdsorg="PO" & right(datei,1) >< ")" , 

            then msg = "Memberangabe fehlt" 

         when syslrecl > 132 then 

            msg = "Satzlänge zu groß" 

         otherwise leave 

      end 

   end 

   else msg='Datei nicht verfügbar' 

   say msg", Eingabe wiederholen." 

end 

say 'Ausgabeklasse angeben (A, R oder X - ENTER=X)' 

parse upper external okl 

if wordpos(okl,"A R X") = 0 then okl='X' 

if okl = "A" then do 

   say "JES-Local-Drucker (ENTER = SPOOL)" 

   parse upper external rmt 

   if length(rmt) > 0 then okl = okl",DEST="rmt 

end 

say 'Druckzeilen je Seite (40 bis 72, ENTER = 60):' 

parse external zeilen 

if zeilen < 40 ! zeilen > 72 then zeilen=60 

say 'Anzahl Exemplare (1 - 99, ENTER = 1):' 

parse external anz_ausg 

if anz_ausg < 1 ! anz_ausg > 99 then anz_ausg = 1 

"newstack" 

name=userid()!!right(time(),1) 

queue "//"name" JOB ("actnum"),NOTIFY="userid() 

queue "/*JOBPARM LINECT="zeilen 

queue "//DRUCK    EXEC PGM=IEBGENER" 

queue "//SYSUT1   DD DSN="datei",DISP=SHR" 

queue "//SYSUT2   DD SYSOUT="okl",COPIES="anz_ausg 

queue "//SYSIN    DD DUMMY" 

queue "//SYSPRINT DD DUMMY" 

queue "XX" 

"submit * end(XX)" 

say 'Job' name 'gestartet um' substr(time(),1,5) 

"delstack" 

end 

exit 

Sehen wir uns das gleiche Beispiel an, bei dem die JCL in einem Variablenstamm vorbereitet wird. 
Der Variablenstamm wird in eine Datei geschrieben. Der Benutzer kann bei Programmende ent-
scheiden, ob die JCL sofort ausgeführt und die Datei wieder gelöscht werden soll, oder ob keine 
sofortige JCL-Ausführung gewünscht ist und die Datei erhalten bleibt. Für dieses Beispiel führen wir 
keine so aufwendige Plausibilitätsprüfung durch. Es geht uns hier nur ums Prinzip: 

/* REXX *  

*/ 

if arg(1) = "T" then trace "?r" 

call msg "OFF" 

say "Abrechnungsnummer (für Job-Account)" 

parse upper external actnum 

say "Druckdatei angeben (ENTER = Ende)" 

parse upper external datei 

if datei="" then exit 

say "Ausgabeklasse angeben (A, R oder X)" 

parse upper external okl 

if okl >< "A" & okl >< "R" then okl = "X" 

if okl = "A" then do 

say "JES-Local-Drucker (ENTER = SPOOL)" 

pull rmt 

if length(rmt) > 0 then okl = okl",DEST="rmt 

end 

say "Druckzeilen je Seite (40 bis 72):" 

parse external zeilen 

if zeilen < 40 ! zeilen > 72 then zeilen=60 

say "Anzahl Ausgaben (bis 99, ENTER = 1)" 

parse external anz_ausg 

if anz_ausg < 1 ! anz_ausg > 99 then anz_ausg = 1 

jname=userid()!!substr(time(),8,1) 

x.1="//"jname" JOB ("actnum"),NOTIFY="userid() 

x.2="/*JOBPARM LINECT="zeilen 

x.3="//DRUCK    EXEC PGM=IEBGENER" 

x.4="//SYSUT1   DD DSN="datei",DISP=SHR" 

x.5="//SYSUT2   DD SYSOUT="klasse",COPIES="anz_ausg 

x.6="//SYSIN    DD DUMMY" 

x.7="//SYSPRINT DD DUMMY" 

if sysdsn(sub.druck)="OK" then "del sub.druck" 

"free fi(jcl)" 

"alloc dd(jcl) da(sub.druck) new unit(sysda) tracks" , 

    "space(1) lrecl(80) recfm(f)" 

"execio * diskw jcl (stem x. finis)" 

say "S (für SUBMIT und Löschen), ENTER (kein SUBMIT)" 

parse upper external antw 

if antw = "S" then do 

   "submit sub.druck" 

   "del sub.druck" 

   say "Job" name "gestartet um" substr(time(),1,5) 

end 

exit 

Soweit ganz einfach. Bauen wir eine JCL-Verarbeitung, bei der die Ausgabe nicht über SPOOL, 
sondern auf Datei laufen soll und nehmen wir zur Demonstration den Sort des MVS. Die Problem-
stellungen vorweg. Die Sort-Ausgabe könnte theoretisch vorhanden sein oder nicht. Abhängig davon 
müssten wir die Datei-Disposition auf der JCL-Seite steuern. Kein Problem, aber: eine bereits vor-
handene Ausgabedatei würde, was Dateiaufbau und -größe angeht, sehr wahrscheinlich nicht ver-
nünftig zu dem Format passen, das wir als Sort-Eingabe nutzen. Sinnvoll ist also, dafür zu sorgen, 
dass die Sort-Ausgabedatei neu erstellt werden kann. Die Datei-Kenngrößen könnten in diesem Fall 
von der Eingabedatei übernommen werden. Die Sortierkriterien sollten ebenfalls flexibel sein. Dem 
Anwender sollte innerhalb gewisser Grenzen überlassen werden, wie viele Sortierfelder er bestim-
men will. Eines ist Pflicht, bis zu vier sind möglich. Dabei können wir nicht davon ausgehen, dass 
der Anwender die Details der Syntax einer SORT FIELD-Anweisung wissen muss. 

/* REXX * JCLSORT  

Eingabedatei variabel, Ausgabe userid.SORTAUS.xxxxxxxx 

Bis zu vier Sortierfelder können definiert werden. 

*/ 

parse upper arg trace . 

if trace="T" then trace '?r' 

else call msg "OFF" 

name_ok='ABCDEFGHIJKLMNOPQRSTUVWXYZ1234567890$§#.()' 

say 'Sort-Eingabedatei vollqualifiziert (ENTER=Ende)' 

do forever 

   parse upper external sort_ein 

   if length(sort_ein)=0 then exit 

   datei=sort_ein 

   call namcheck 

   if fehl = 0 then do 

      if sysdsn("'"sort_ein"'") >< 'OK' then do 

         msg='Datei oder Member nicht verfügbar' 

         fehl=1 

      end 

      else do 

         call listdsi "'"sort_ein"'" 

         fehl=1 

         select 

            when right(sort_ein,1) >< ")" & , 

               sysdsorg="PO" then  

                  msg='Memberangabe fehlt' 

            when wordpos(sysdsorg,"PS PO") = 0 then 

               msg="Falsche Dateiorganisation" 

            otherwise fehl=0 

         end 

      end 

   end 

   if fehl=0 then leave 

   say msg ', bitte nochmal' 

end 

say 'Letzten Qualifier der Ausgabedatei' 

say userid()'.SORTAUS.???????? angeben' 

parse upper external sort_aus . 

name_ok=substr(name_ok,1,length(name_ok)-2) 

datei=sort_aus 

call namcheck 

if fehl=1 ! length(sort_aus) = 0 ! , 

   datatype(left(aus,1),u)=0 ! , 

   length(sort_aus) > 8 , 

   then do 

      say 'Name formal falsch, STANDARD wird benutzt' 

      aus='STANDARD' 

   end 

   if sysdsn('sortaus.'sort_aus)='OK' then 

   "del sortaus."aus 

   x=1 

   s_folge="" 

   do forever    /* abholen der Sortierkriterien */ 

      say "Start Sortierfeld" x "(1 bis" syslrecl")" 

      if x > 1 then say "(ENTER = Ende Abfrage)" 

      do forever 

         parse external von . 

         select 

            when length(von)=0 & x > 1 then do 

               krit='ENDE' 

               leave 

            end 

            when datatype(von) ^= 'NUM' then 

               msg='Eingabe nicht numerisch' 

            when von < 1 ! von > syslrecl then 

               msg='Beginnposition falsch' 

            otherwise leave 

         end 

      end 

      if krit='ENDE' then leave 

      if von=syslrecl then do 

      say 'Beginn=Satzende, Länge wird auf 1 gesetzt' 

      lang=1 

   end 

   else do 

      max=syslrecl-von+1 

      say 'Länge des' x'. Sortierfeldes, max.' max 

      parse external lang . 

      if lang < 1 ! lang > max then do 

         say 'Falsche Länge, 1 wird angenommen' 

         lang=1 

      end 

   end 

   do until richt = 'A' ! richt = 'D' 

      say 'Sortierfolge: A=Aufsteigend, D=Absteigend' 

      parse upper external richt . 

   end 

   s_folge=s_folge','von','lang',CH,'richt 

   x=x+1 

   if x > 4 then leave 

end 

select 

   when sysunits='CYLINDER' then unit='CYL' 

   when sysunits='TRACK' then unit='TRK' 

   otherwise unit=sysblksize 

end 

pri=sysprimary 

sec=sysseconds 

s_folge=space(substr(s_folge,2,length(s_folge)),0) 

queue "//SORT    EXEC PGM=ICEMAN" 

queue "//SORTIN  DD DSN="sort_ein",DISP=SHR" 

queue "//SORTOUT DD DSN="userid()'.SORTAUS.'aus"," 

queue "//           DISP=(NEW,CATLG,DELETE)," 

queue "//           UNIT=SYSDA,VOL=SER="sysvolume"," 

queue "//           SPACE=("unit",("pri","sec"))" 

do cnt=1 to 8 

   queue "//SORTWK0"cnt "DD UNIT=SYSDA," 

   queue "//            SPACE=("unit",("pri","sec"))" 

end 

queue "//SYSOUT   DD SYSOUT=*" 

queue "//SYSIN    DD *" 

queue " SORT FIELDS=("s_folge")" 

queue "XX" 

j=substr("ABCDEFGHIJKLMNOPQRSTUVWXYZ",random(1,26),1) 

"submit * end(XX) jobchar("j")" 

exit 

 

/* SUBROUTINE NAMCHECK * Prüft DSN auf Sonderzeichen */ 

namcheck: 

fehl=0 

do x=1 to length(datei) 

   if pos(substr(datei,x,1),name_ok)=0 then do 

      msg='Nur A-Z  0-9  $  #  (  ) und  .' 

      fehl=1 

      leave 

   end 

end 

return 

1.29.14. Strukturprüfung 

Wichtig für die Lesbarkeit eines Programmes ist neben einer vernünftigen Logik auch die Schreib-
weise. Konsequentes Ein- und Ausrücken der Befehle gehört ebenso dazu, wie auf die Unart zu 
verzichten, mehrere unterschiedliche Befehle in eine Zeile zu kodieren. Wird ein Programm spalten-
gerade mit allen seinen Befehlen ab Spalte 1 kodiert, ist es sehr schwierig, logische Abhängigkeiten 
zu erkennen, ein fehlendes oder überflüssiges END zu finden und den Verlauf des Programmes am 
Terminal zu verfolgen (ich meine hier natürlich nicht im Ablauf unter TRACE). Werden völlig unmo-
tivierte Ein- und Ausrückungen gemacht, wird das Ganze noch verrückter. 

Im Anschluss deshalb ein Versuch, die unsaubere Schreibweise, sowie zu viele oder zu wenige END 
von einem REXX-Programm überprüfen zu lassen. Das Ergebnis wird in eine sequentielle Datei 
geschrieben und (wir nehmen uns einfach mal eine kleine Anleihe beim Dialog Manager) durch 
Browse gezeigt. 

Dass das Strukturierprogramm keinen lauffähigen Quellencode erzeugt, ist Absicht (sonst würde 
sich niemand mehr Mühe geben, saubere Programme zu schreiben). Dieses Manko (sofern man es 
so sieht) kann aber mit ein paar kleinen Änderungen behoben werden. 

/* REXX  * RXSTRU  

*/ 

arg source ein_aus . 

if ein_aus < 2 ! ein_aus > 4 then ein_aus = 3 

 

pref="!"copies(" ",ein_aus-1) 

say center(" REXX-Quelltext strukturieren ",79,"*") 

do until sysdsn("'"source"'") = "OK" & right(source,1)=")" 

   say "dateiname(programm) vollqualifiziert eingeben" 

   pull source 

end 

"alloc dd(source) da('"source"') shr reuse" 

if rc > 0 then do 

   say "Alloc verunglückt mit RC="rc 

   exit 

end 

"newstack" 

"execio * diskr source (finis" 

z = 0 

do queued() 

   parse pull satz 

   satz = strip(satz,"B") 

   pos = pos(";",satz) 

   do while pos > 0 & pos >< lastpos(";",satz) 

      z = z+1 

      qt.z = substr(satz,1,pos) 

      satz = substr(satz,pos+1,length(satz)-pos) 

      pos = pos(";",satz) 

   end 

   z = z+1 

   qt.z = satz 

end 

qt.0 = z 

spalte = 0 

anz_ein = 0 

anz_aus = 0 

qtn.1  = "REXX" source "von" userid() date() time() 

do z = 2 to qt.0+1 

   qt.such = qt.z 

   pos_cmta = pos("/*",qt.such) 

   pos_cmte = pos("*/",qt.such) 

   if pos_cmta > 0 then do 

      qt.such = left(qt.z,pos_cmta-1) 

   end 

   select 

      when pos("DO",translate(qt.such))  > 0 then do 

         qtn.z = copies(pref,spalte/ein_aus)!!qt.z 

         spalte = spalte + ein_aus 

         anz_ein = anz_ein + 1 

      end 

      when pos("SELECT",translate(qt.such))>0 then do 

         qtn.z = copies(pref,spalte/ein_aus)!!qt.z 

         spalte = spalte + ein_aus 

         anz_ein = anz_ein + 1 

      end 

      when pos("END",translate(qt.such)) > 0 then do 

         if spalte >= ein_aus then do 

            spalte = spalte - ein_aus 

         end 

         qtn.z = copies(pref,spalte/ein_aus)!!qt.z 

         anz_aus = anz_aus + 1 

      end 

      otherwise do 

         qtn.z=copies(pref,spalte/ein_aus)!!qt.z 

      end 

   end 

end 

z = qt.0 + 1 

qtn.z = ">>> Anzahl DOs/SELECTs.:" anz_ein 

z = z + 1 

qtn.z = ">>> Anzahl ENDs........:" anz_aus 

"del rexx.stru" 

"alloc dd(aus) da(rexx.stru) new" , 

   "unit(sysda) space(1) tracks dsorg(ps)" , 

   "lrecl(80) blksize(80) recfm(v)" 

"execio * diskw aus (stem qtn. finis" 

"free dd(source aus)" 

if sysvar(sysispf) = "ACTIVE" then 

   address ispexec "edit dataset(rexx.stru)" 

exit 

1.29.15. Sortieren 

Wie versprochen, möchte ich nun an dieser Stelle zeigen, wie eine kleine Sortier-Routine aussehen 
kann. Grundsätzlich wollen wir die Problemlösung als UDF (User Defined Function) anlegen, um sie 
in jedem beliebigen Programm einsetzen zu können. Die Funktion übernimmt eine Wortkette und 
gibt sie sortiert wieder zurück. Selbstverständlich könnten wir die Funktion über einen weiteren Pa-
rameter veranlassen, ein absteigendes Ergebnis zu liefern, aber ein bisschen Kreatives möchte ich 
gerne für euch übriglassen (Bescheidenheit war immer meine Stärke). Es bedarf nur ein paar kleiner 
Veränderungen, viel Spaß dabei. 

/* REXX * SORT  

*/ 

arg parm 

do while length(parm) > 0 

   done=0 

   do pos = 1 to words(parm) -1 

      idx1 = word(parm,pos) 

      idx2 = word(parm,pos+1) 

      if idx2 > idx1 then do 

         parm = subword(parm,1,pos-1) IDX2 IDX1 , 

         subword(parm,pos+2) 

         done=1 

      end 

   end 

   if done=0 then leave 

end 

return parm 

Sehen wir uns das gleiche Beispiel leicht geändert an: 

/* REXX * SORT */ 

arg parm 

ende=words(parm) 

do x=1 to ende 

   parm.x=word(parm,x) 

end 

tmp=ende+1 

parm.tmp="ff"x 

do while ende > 0 

   done=0 

   do x1 = 1 to ende 

      x2=x1+1 

      if parm.x1 > parm.x2 then do 

         help = parm.x1 

         parm.x1 = parm.x2 

         parm.x2 = help 

         done=1 

      end 

   end 

   if done = 0 then do 

      parm="" 

      do x=1 to end 

         parm = parm parm.x 

      end 

      leave 

   end 

end 

return strip(parm) 

Zunächst erscheint die zweite Variante viel aufwändiger als die erste. Dies ist bedingt richtig. Natür-
lich ist es mehr Arbeit, die gelieferte Wortfolge in einen Variablenstamm umzubauen und nach er-
folgter Sortierung wieder zurück zu verwandeln. Der Vorteil liegt aber darin, dass innerhalb der Ver-
arbeitungsschleife mit Compound-Variablen wesentlich schneller gearbeitet werden kann als mit den 
WORD-Funktionen im ersten Beispiel. Diesen Vorteil registriert man allerdings erst, wenn die Länge 
der angelieferten Wortkette über ein bestimmtes Maß hinausgeht. Werden in der Wortkette unter 
fünf Werte geliefert, kann das zweite Beispiel sogar mehr Zeit brauchen als das erste.  

Deutlich optimiert wird die Sort-Variante auch Bubble-Sort genannt. Hier haben wir nicht schier end-
lose Durchläufe, immer in der gleichen Richtung, bis in einem Durchlauf keine Änderung mehr zu 
machen war. Wir durchlaufen die Wortkette einmal von links nach rechts. Stellen wir beim Vergleich 
zweier Werte fest, dass der rechte Wert kleiner dem linken ist, drehen wir die beiden Werte und 
halten fest, wieweit wir auf dem Weg nach rechts bereits waren. Anschließend gehen wir mit dem 
ausgetauschten Wert sofort nach links und schieben ihn so weit in Richtung Anfang der Wortkette, 
wie dies nötig ist. Anschließend können wir unsere Verarbeitung nach rechts an der unterbrochenen 
Stelle fortsetzen. 

/* REXX * BSORT */ 

arg parm 

parm.0="00"x 

ende=words(parm) 

do x=1 to end 

   parm.x=word(parm,x) 

end 

tmp=ende+1 

parm.tmp="ff"x 

do x1 = 1 to words(parm) -1 

   x2=x1+1 

   if parm.x1 > parm.x2 then do 

      help = parm.x1 

      parm.x1 = parm.x2 

      parm.x2 = help 

      do y1 = x1 to 1 by -1 

         y2=y1-1 

         if parm.y1 >= parm.y2 then leave 

         help = parm.y1 

         parm.y1 = parm.y2 

         parm.y2 = help 

      end 

   end 

end 

parm="" 

do x=1 to ende 

   parm = parm parm.x 

end 

return strip(parm) 

Auch dieses Beispiel mag bei erster Betrachtung einen eher komplizierten Eindruck vermitteln, als 
das erste. Interessant ist aber das Zeitverhalten der drei unterschiedlichen Varianten im Ablauf. Um 
keine störenden Nebenwirkungen zu haben, wieder ein Testergebnis auf einem PC. Sortiert wurde 
jeweils eine Zahlenkette, die über einen Zufallsgenerator zusammengebaut wurde. Die unterschied-
lichen Ergebnisse beruhen auf dem Aufbau der Kette (Je nachdem, wie viele Umstellungen für die 
richtige Sortierung nötig waren, wird, je weniger oft dies der Fall ist, die erste Lösung zeitlich etwas 
profitieren): 

Durchläufe Zeitbedarf in Sekunden 

Variante 1 Variante 2 Variante 3 

1. Durchlauf 24,40 13,95 5,49 

2. Durchlauf 25,54 12,63 5,33 

3. Durchlauf 27,02 13,40 5,66 

4. Durchlauf 27,29 13,78 5,39 

5. Durchlauf 24,33 11,92 5,55 

 

Abschließend zwei Beispiele für die Nutzung als Funktion oder Subroutine in Ermangelung einer 
Built-in-Sort-Funktion. 

Das erste Beispiel kann benutzt werden, um eine übergebene Wortkette zu sortieren. Das zweite 
sortiert den Inhalt des Stackbereiches. Hier wäre vorstellbar, die Sortparameter als Argumente zu 
übergeben und nicht immer ab Stelle 1 zu sortieren. Viel Spaß bei der Realisierung. 

/* REXX * SORTWORD 

   Zweck....: Übernimmt eine Wortkette und gibt 

              sie aufsteigend sortiert zurück 

  

   Erstellt.: 09.2008 Wittemann 

  

*/ 

parse arg parm 

parm = "00"x parm "FF"x 

say parm 

DO §V = 1 TO words(parm)-1 

   IF word(parm,§V) > word(parm,§V+1) THEN DO 

      parm =  subword(parm,1,§V-1) , 

              word(parm,§V+1) word(parm,§V) , 

              subword(parm,§V+2) 

      DO §Z = §V TO 2 by -1 

         IF word(parm,§Z) < word(parm,§Z-1) THEN DO 

            parm =  subword(parm,1,§Z-2) , 

                    word(parm,§Z) word(parm,§Z-1) , 

                    subword(parm,§Z+1) 

         END 

         else leave 

      END 

      done=1 

   END 

END 

RETURN strip(subword(parm,2,words(parm)-2)) 

 

 

/* REXX * SORTSTCK 

 

   Zweck..: Liest Sätze im Stack und stellt sie sortiert zurück 

   PARMs..: §S = Beginnposition Sortierfeld 

            §L = Länge Sortierfeld 

 

   Erstellt 09.2008 Wittemann 

*/ 

if queued() = 0 then return 0 

parse source . rxTYP . 

if rxTYP = "FUNCTION" then arg §S,§L,DUMMY 

else                       arg §S §L . 

if §S = "" then §S = 1 

if §L = "" then §L = 10 

do §=1 to queued() 

   parse pull satz.§ 

end 

satz.0 = "00"x 

satz.§ = "FF"x 

DO §V = 1 TO §-1 

   §Next = §V+1 

   IF substr(satz.§V,§S,§L) > , 

      substr(satz.§Next,§S,§L) THEN DO 

      satz.§V    = bitxor(satz.§V,satz.§Next) 

      satz.§Next = bitxor(satz.§Next,satz.§V) 

      satz.§V    = bitxor(satz.§V,satz.§Next) 

      DO §Z = §V TO 1 by -1 

         §Prev = §Z-1 

         IF substr(satz.§Z,§S,§L) < , 

            substr(satz.§Prev,§S,§L) THEN DO 

            satz.§Z    = bitxor(satz.§Z,satz.§Prev) 

            satz.§Prev = bitxor(satz.§Prev,satz.§Z) 

            satz.§Z    = bitxor(satz.§Z,satz.§Prev) 

         END 

         else leave 

      END 

   END 

END 

do §=1 to §-1 

   queue satz.§ 

end 

RETURN 0 

Und wie macht's der Profi? 

/* REXX  

   Sortieren eines Stems via BPXWUNIX  

   Gefunden bei John McKown (REXX-TSO Group)  

*/ 

_=0 

_=_+1;A._="Katze" 

_=_+1;A._="Salamander" 

_=_+1;A._="Hund" 

_=_+1;A._="Vogel" 

_=_+1;A._="Maus" 

_=_+1;A._="Pferd" 

_=_+1;A._="Aligator" 

_=_+1;A._="Krokodil" 

A.0=_ 

stdout.0=0 

stderr.0=0 

call bpxwunix "/bin/sort",a.,b.,stderr. 

say;say left("Stemsort CHAR",50,"-") "Elemente:" a.0;say 

say "Nr. Unsorted (Stem A.)-- Sorted (Stem B.)----" 

do  i=1 to a.0 

    say right(i,3) left(a.i,20) b.i 

end 

say;say 'stderr.0='stderr.0 

do  i=1 to stderr.0 

    say "stderr."i"="stderr.i 

end 

 

_=0 

_=_+1;A._="7" 

_=_+1;A._="28" 

_=_+1;A._="3" 

_=_+1;A._="105.9" 

_=_+1;A._="88" 

_=_+1;A._="307" 

_=_+1;A._="2" 

_=_+1;A._=random() 

_=_+1;A._=random() 

_=_+1;A._=random() 

_=_+1;A._=random() 

A.0=_ 

stdout.0=0 

stderr.0=0 

call bpxwunix "/bin/sort -n",a.,a.,stderr. 

say;say left("Stemsort NUM",50,"-") "Elemente:" a.0;say 

do  i=1 to a.0 

    say left("a."i,6) "=" a.i 

end 

say;say 'stderr.0='stderr.0 

do  i=1 to stderr.0 

    say "stderr."i"="stderr.i 

end 

 

Es ist nicht immer sinnvoll, das Rad neu zu erfinden. Der Sort aus Unix ist in der Lage, einen Stem 
"in Place" zu sortieren oder das sortierte Ergebnis in einen anderen Stem zu schreiben. 

Eine weitere Variante bestünde darin, eine temporäre ISPF-Tabelle zu erstellen, diese zu sortieren 
und die unsortierten Daten in die Tabelle zu schreiben und danach wieder auszulesen. Würde man 
ein bestimmtes Feld als Schlüssel definieren, wären damit auch doppelte Einträge eliminiert. Dazu 
findet sich ein Beispiel im Kapitel ISPF. 

1.29.16. Druckaufbereitung / Rechenfähigkeit 

Wenn man Cobol kennt, schätzt man die Möglichkeit, Felder als druckaufbereitet oder Rechenfähig 
zu deklarieren. Das folgende Programm erlaubt die Konvertierung vom rechenfähigen ins druckauf-
bereitete Format und zurück. 

/* REXX * §FMT 

 

Ohne Parameter (D=> Druckaufbereitet, R=> Rechenfähig) 

wird Druckaufbereitet angenommen. 

*/ 

parse source . RxTyp RxName .                                                    

if RxTyp = "FUNCTION" then arg input,fmt,vor,nach,dummy                          

else                       arg input fmt vor nach .                              

if fmt = "D" then return Druck_Aufbereitet()                                     

else              return Rechenfaehig()                                          

exit                                                                             

                                                                                 

Druck_Aufbereitet:                                                               

   parse var input int"."dec                                                     

   Druck = reverse(strip(space( ,                                                

           translate("abc.def.ghi.jkl.mno.pqr.stu.vwx.yz", ,                     

           reverse(int),"abcdefghijklmnopqrstuvwxyz") ,                          

           ,0),,"."))                                                            

   if datatype(vor,"W") then Druck = right(druck,vor)                            

   if ^datatype(nach,"W") then nach = length(dec)                                

                                                                                 

   if dec <> "" then Druck = Druck","left(dec,nach,0)                            

return Druck                                                                     

                                                                                 

Rechenfaehig:                                                                    

   parse var input int","dec                                                     

   num = space(translate(int,,"."),0)                                            

   if datatype(vor,"W") then num = right(num,vor,0)                              

   if ^datatype(nach,"W") then nach = length(dec)                                

   if dec <> "" then num = num"."left(dec,nach,0)                                

return num 

1.29.17. Datum und Zeit 

Hier einige kleine Helferlein für all jene, die sich immer wieder mit Datumsvariablen in jeder denkbaren Form 
herumschlagen müssen. Vielfach muss ein Datum auf formale Richtigkeit überprüft werden. Praktisch wäre 
oft, mit Datumsfeldern rechnen zu können (Eine beliebige Anzahl Tage aufrechnen oder abziehen und das 
Ergebnis sollte wieder ein gültiges Datum sein). Die Wochentage eines bestimmten Datums wären oft auch 
nicht schlecht. Und zur Krönung: Die Ermittlung der beweglichen Feiertage (Alle kirchlichen beweglichen Fei-
ertage sind abhängig von Ostern. Ostern wiederum ist abhängig vom ersten Vollmond im Frühling. Dieser 
berechnet sich nach einer Formel von Karl Friedrich Gauß. Wir wollen hier nur die Formel umsetzen). 

Die Problemlösungen sind als UDFs (benutzerdefinierte Funktionen) geschrieben und müssen lediglich inner-
halb der SYSPROC/SYSEXEC-Verkettung gefunden werden können. 

1.29.17.1. Rechnen mit einem Datum 

/* REXX * DATCNT 

Input: Datum im Format tt.mm.jjjj*/ 

arg datum,tage 

parse var datum tt "." mm "." jjjj 

base = date("B",jjjj !! mm !! tt,S) + tage 

return translate("Tt.Mm.CcJj",date("S",base,"B"),"CcJjMmTt") 

1.29.17.2. Tage zwischen 2 Datumsangaben 

/* REXX * DATEDIFF * FUNCTION 

Eingabe DATUM1 DATUM2 im Format tt.mm.jjjj*/ 

arg datum1, datum2 

return basedat(datum2) – basedat(datum1) 

basedat: procedure 

parse value arg(1) with tt "." mm "." jjjj 

return date("B",jjjj !! mm !! tt,"S") 

1.29.17.3. Kalenderwoche 

/* REXX * WEEK * FUNCTION 

   Ermittelt die Kalenderwoche zu einem Datum nach  

   europäischer Norm (die Woche beginnt mit einem  

   Montag). 

   Ohne ARG() wird DATE() angenommen. 

*/ 

parse source . rxTYP rxNAME . 

arg datum 

if datum = "" then  

   datum = Translate("Dd.Mm.CcYy",Date("S"),"CcYyMmDd") 

parse var datum tt "." mm "." jjjj 

dat_b = date("B",jjjj !! mm !! tt,"S") 

next_sunday = dat_b + (6 - dat_b // 7) 

if mdw1(jjjj+1) <= dat_b then do 

   kw = 1 

   jjjj = jjjj + 1 

end 

else do 

   mon_week_1 = MDW1(jjjj) 

   if mon_week_1 > dat_b then do 

      jjjj = jjjj - 1 

      mon_week_1 = MDW1(jjjj) 

   end 

   kw = (next_sunday+1 - mon_week_1) % 7 

end 

if rxTYP <> "COMMAND" then return kw jjjj 

say date("W",dat_b,"B")"," datum ">> KW" kw "("jjjj")" 

exit 

  

MDW1: procedure 

   datum = date("B",arg(1)"0101","S") 

   tag = datum // 7 + 1 

   if tag <= 4 then _1st_of_week_1 = datum - tag+1 

   else             _1st_of_week_1 = datum + (7 -tag+1) 

return _1st_of_week_1 

1.29.17.4. Ostersonntag 

/* REXX * Beschreibung am Ende 

*/ 

parse source . RxTyp RxName . 

if RxTyp = "FUNCTION" then arg jahr,format,dymmy 

else                       arg jahr format . 

if jahr < 1582 then jahr = left(date("S"),4) 

a=jahr//19 

b=jahr//4 

c=jahr//7 

d=((19*a+24)//30) 

e=((2*b+4*c+6*d+5)//7) 

f=22+d+e 

if f = 56 & d=28 & e=6 & a > 10 then f=49 

if f <= 31 then m="03" 

else do 

   f=f-31 

   m="04" 

end 

call convert jahr !! m !! right(f,2,0) 

if RxTyp <> "COMMAND" then return result 

say result 

exit 0 

 

convert: 

   select 

      when format = "J" then oso = julian(arg(1)) 

      when format = "E" then oso = date("E",arg(1),"S") 

      when format = "N" then oso = date("N",arg(1),"S") 

      when format = "B" then oso = date("B",arg(1),"S") 

      when format = "O" then oso = date("O",arg(1),"S") 

      when format = "U" then oso = date("U",arg(1),"S") 

      when format = "M" then oso = word("März April",m-2) 

      when format = "D" then oso = right(julian(arg(1)),3) 

      otherwise oso = translate("Tt.Mm.CcJj",arg(1),"CcJjMmTt") 

   end 

return oso 

 

julian: procedure 

   arg 1 j 5 m 7 t 

   tim = "31" 28+(j//4=0)-(j//100=0)+(j//400=0) , 

         "31 30 31 30 31 31 30 31 30 31" 

   do i=1 to m-1 

      t = t + word(tim,i) 

   end 

return j !! right(t,3,0) 

 

/* Programmbeschreibung *********************************************** 

 

Ermittelt das Datum des Ostersonntags für ein beliebiges Jahr nach 

Der Formels von Carl Friedrich Gauß 

 

Parameter 1: Jahr. Wenn fehlt oder < 1582 wird das aktuelle angenommen 

Parameter 2: Format. Abhängig vom Wert wird zurückgegeben 

             J (Julean)    jjjjttt 

             E (European)  tt/mm/jj 

             N (Normal)    tt Mmm jjjj 

             B (Base) 

             O (Ordered)   jj/mm/tt 

             U (USA)       mm/tt/jj 

             M (Month)     März | April 

             S (Sorted)    jjjjmmtt 

             ? (Ansonsten) tt.mm.jjjj 

*/ 

1.29.17.5. Deutscher Wochentag 

/* REXX §DOWD 

Ermittelt den deutschen Wochentag eines Datums im Format tt.mm.jjjj.  

Wird kein Datum übergeben, wird das Tagesdatum angenommen 

*/ 

Arg datum 

If datum = "" then datum = translate("Tt.Mm.CcJj",date("S"),"CcJjMmTt") 

parse var datum tt "." mm "." jjjj 

dow = date("W",jjjj !! mm !! tt,"S") 

days = "Monday Montag" , 

       "Tuesday Dienstag" , 

       "Wednesday Mittwoch" ,  

       "Thursday Donnerstag" , 

       "Friday Freitag" , 

       "Saturday Samstag" , 

       "Sunday Sonntag" 

Return word(days,wordpos(dow,days)+1) 

1.29.17.6. Feiertage 

/* REXX * FUNCTION * FTAGE 

   Feiertagsberechnung "Sonderanfertigung Schweiz" (mit 

   "Knabenschießen", "Sechseläuten", dem Schweizer  

   Berchtoldstag und dem Nationalfeiertag). Das Programm  

   läuft unter ISPF und prüft zu Beginn, ob die Feiertage  

   für das entsprechende Jahr bereits im Profile-Pool  

   abgelegt wurden. Wenn nicht, werden sie berechnet und  

   in den Profile-Pool geschrieben. 

   Die externe Funktion §OSO wird benutzt. 

*/ 

SIGNAL ON SYNTAX NAME fehler 

address ispexec 

"control errors return" 

PARSE SOURCE . prog_typ . 

ARG jahr 

IF LENGTH(jahr) = 0 THEN jahr = WORD(DATE(),3) 

savejahr = "FT"jahr 

"vget (&savejahr) profile" 

if value(savejahr) = "" then do 

   parse value §oso(jahr) with tt "." mm "." jjjj 

   osobase = date("B",jjjj !! mm !! tt,"S") 

   ft_date = "01.01."jjjj "Neujahr" , 

             "02.01."jjjj "Berchtoldstag_(CH)" , 

             "06.01."jjjj "3_König" , 

             !! "" 

   karfr   = datum(osobase - 2) 

   oso     = datum(osobase) 

   omo     = datum(osobase + 1) 

   do z=15 to 21 /* Sechseläuten Zürich */ 

      sechse=RIGHT(z,2,0)'.04.'jahr 

      IF §dowd(sechse) = 'Montag' THEN LEAVE 

   end 

   parse var sechse tt "." mm "." jjjj 

   sechseb = date("B",jjjj !! mm !! tt,"S") 

   SELECT 

      WHEN sechseb=osobase+1 THEN sechse=datum(sechseb+7) 

      WHEN sechseb=osobase-6 THEN sechse=datum(sechseb-7) 

      OTHERWISE sechse = datum(sechseb) 

   END 

   if sechseb < osobase then do 

      ft_date = ft_date , 

                sechse "Sechseläuten_(Zürich)" , 

                karfr  "Karfreitag" , 

                oso    "Ostersonntag" , 

                omo    "Ostermontag" , 

                !! "" 

   end 

   else do 

      ft_date = ft_date , 

                karfr  "Karfreitag" , 

                oso    "Ostersonntag" , 

                omo    "Ostermontag" , 

                sechse "Sechseläuten_(Zürich)" , 

                !! "" 

   end 

   ft_date = ft_date , 

             "01.05."jjjj      "Tag_der_Arbeit" , 

             datum(osobase+39) "Jesus_Skydrive" , 

             datum(osobase+49) "Pfingstsonntag" , 

             datum(osobase+50) "Pfingstmontag"  , 

             !! "" 

   DO x=1 TO 7 /* Knabenschießen Zürich */ 

      boybase = date("B",jahr"09"right(x,2,0),"S") 

      boydate = datum(boybase) 

      IF §dowd(boydate) = 'Sonntag' then do 

         boydate = datum(boybase+8) 

         LEAVE 

      END 

   END 

   ft_date = ft_date , 

             "01.08."jjjj "Nationalfeiertag_(CH)" , 

             boydate      "Knabenschießen_(Zürich)" , 

             "15.08."jjjj "Marys_Skydrive" , 

             "03.10."jjjj "Tag_der_Einheit" , 

             "25.12."jjjj "1.Weihnachtstag" , 

             "26.12."jjjj "2.Weihnachtstag" , 

             !! "" 

   interpret savejahr "= ft_date" 

   "vput (&savejahr) profile" 

END 

dat = value(savejahr) 

ftg = wordpos(datum(date("B")),value(savejahr)) 

IF prog_typ <> "COMMAND" THEN return (ftg > 0) dat 

SAY "Feiertage für das Jahr:" jahr;SAY 

if ftg > 0 then  

   say "Heute ist Feiertag ("word(dat,ftg+1)")" 

say 

DO i=1 TO words(dat) by 2 

   say i ">>" WORD(dat,i) WORD(dat,i+1) 

   IF i // 20 = 0 THEN pull 

END 

EXIT 

datum: procedure 

return , 

  Translate("Dd.Mm.CcYy",date("S",arg(1),"B"),"CcYyMmDd") 

FEHLER: 

   SAY "RC.:" RIGHT(rc,4)   ">>" ERRORTEXT(rc) 

   SAY RIGHT(sigl,9) ">>" SOURCELINE(sigl) 

   TRACE "?R" 

   NOP 

EXIT 

1.29.17.7. Datum formal richtig 

Es gibt viele Möglichkeiten, ein Datum auf Plausibilität zu prüfen. Wie aber macht es ein REXX-
Programmierer? Er überlässt die Arbeit dem Interpreter. Man versucht das Datum in ein beliebiges 
anderes Format zu konvertieren. Tritt ein Syntaxfehler auf, war das Eingangsdatum formal falsch. 

/* REXX 

   Input: Datum imn Format tt.mm.jjjj 

   Rückgabe: 1 > Datum korrekt 

             0 > Datum inkorrekt 

*/ 

arg tt "." mm "." jjjj 

signal on syntax 

Erg = 1 

base = date("B",jjjj !! mm !! tt,"S") 

Continue: 

Return Erg 

 

syntax: 

   Erg = 0 

   signal Continue 

1.29.17.8. Ist ein Datum (tt.mm.jjjj) ein Monatsletzter? 

Man kann es auskodieren, aber den letzten Tag eines Monats hat man auch, wenn man 1 auf das 
Datum addiert und sich im Ergebnis der Monat ändert. 

/* REXX * ULTCHK * Prüft, ob ein Datum ein Monatsletzter ist */ 

arg tt "." mm "." jjjj 

msg = tt"."mm"."jjjj "ist kein Ultimo" 

if substr(date("S",date("B",jjjj !! mm !! tt,"S")+1,"B"),5,2) <> mm , 

   then msg = delstr(msg,16,1) 

say msg 

exit 

1.29.17.9. DS-Qualifier mit Datum und Uhrzeit 

Wenn ein Zeitstempel in einem Dateinamen gebraucht wird, der nicht länger als 8 Byte ist und bis 
zur hundertstel Sekunde geht, wird es manchmal "schwierig" (alles ist einfach, wenn man weiß, wie 
es geht). 

/* Rexx 

   Programmbeschreibung und Update History am Ende 

*/ 

parse source . RxTyp RxName . 

parse upper arg Input 

signal on syntax name Invalid_Input 

 

if Input = "" then Input = date("S") time("L") 

parse value space(translate(Input,,"/.:-"),0) with input 

 

numeric digits 50 

BaseDays = date("B","19640101","S")-1 

OkChars  = "ABCDEFGHIJKLMNOPQRSTUVWXYZ§$#0123456789" 

 

Select 

   when datatype(Input)       = "NUM" then Ergebnis = DateQualifier() 

   when verify(Input,OkChars) = 0     then Ergebnis = Date_Time() 

   otherwise do 

      ZEdLMsg = "DateQualifier enthält ungültige Zeichen '"Input"'" 

      Ergebnis = "-Error" 

   End 

End 

 

if ZEdLMsg = "ZEDLMSG" then ZEdLMsg = Ergebnis 

Back_from_Error: 

if RxTyp <> "COMMAND" then return Ergebnis 

if sysvar("SYSISPF")="ACTIVE" then address ispexec "Setmsg msg(isrz000)" 

else say Ergebnis 

exit 

 

DateQualifier: procedure expose BaseDays Input OkChars ZEdLMsg 

   call time "R";do while time("E") < .01;end 

   if left(Input,4) < 1964 ! left(Input,4) > 3224 then do 

      ZEdLMsg  = "<"Input"> liegt auserhalb des erlaubten Bereiches", 

                 "(1964-3224)" 

      return "-Error" 

   end 

 

   Input = overlay(Input,"1964010100000000",1,length(Input)) 

   parse var Input 1 Datum 9 Std 11 Min 13 Sec 15 mSec 17 . 

 

   milliSec = ((date("B",Datum,"S")-BaseDays) * 60 *60 * 24, 

              + Std * 3600 + Min * 60 + Sec)*100 + mSec 

   DateQualifier = "" 

 

   do § = 1 until milliSec = 0 

      Rest          = milliSec // length(OkChars) + 1 

      milliSec      = milliSec %  length(OkChars) 

      DateQualifier = substr(OkChars,Rest,1) !! DateQualifier 

   End 

return right(DateQualifier,8,"A") 

 

Date_Time: procedure expose BaseDays Input OkChars 

   milliSec = 0 

   Input    = reverse(Input) 

 

   do §=1 to Length(Input) 

      milliSec = milliSec + (pos(substr(Input,§,1),OkChars)-1), 

                 * length(OkChars) ** (§-1) 

   End 

 

   milliSec = milliSec + BaseDays * 24 * 60 * 60 * 100 

 

   mSec     = right(milliSec // 100,2,0) 

   Sekunden = (milliSec - mSec) / 100 

   Sec      = right(Sekunden // 60,2,0) 

   Sekunden = Sekunden – Sec 

   Min      = right((Sekunden // 3600) / 60,2,0) 

   Sekunden = Sekunden - Min * 60 

   Std      = right((Sekunden // (3600 * 24)) / 3600,2,0) 

   Tage     = (Sekunden - Std * 3600) / (3600 * 24) 

 

   Date_Time = translate("Tt.Mm.CcJj",date("S",Tage,"B"),"CcJjMmTt") 

   if Std + Min + Sec + mSec > 0 then do 

      Date_Time = Date_Time Std":"Min":"Sec"."mSec 

   End 

return Date_Time 

 

Invalid_Input: 

   /****************************************************************** 

   * Datum ist auserhalb des erlaubten Bereiches                     * 

   * 01.01.1964 - 10.02.3225                                         * 

   ******************************************************************/ 

   ZEdLMsg  = "" 

   Result   = "-Error" 

   signal Back_from_Error 

exit 

 

/* Programmbeschreibung ************************************************ 

 

   Erstellt ein DS-Teil-Qualifier aus Datum Zeit Millisekunde oder 

   Wandelt DS-Teil-Qualifier wieder in Datum Zeit Millisekunde um 

   Als Command oder als Funktion nutzbar 

 

   Aufrufformat: 

    >-DATEQUAL----------------------------------------------->< 

                |                                          | 

                |-yyyy-------------------------------------| 

                |      |                                  || 

                |       -mm------------------------------- | 

                |          |                             | | 

                |           -dd--------------------------  | 

                |               |                       |  | 

                |                -hh--------------------   | 

                |                    |                 |   | 

                |                     -mm--------------    | 

                |                         |           |    | 

                |                          -ss--------     | 

                |                              |     |     | 

                |                               -hs--      | 

                 -ds-teil-qualifier------------------------  

 

   Wird kein Parameter angegeben, wird das aktuelle Datum und die 

   aktuelle Zeit inklusive der hundertstel Sekunde angenommen. 

 

-- Update History ------------------------------------------------------ 

   05.07.2016 Zitzmann, Erstellt 

*/ 

1.29.18. TSO Console Command 

Für die Ausführung von MVS- oder JES-Commands steht die Möglichkeit zur Nutzung einer MVS-
Konsole zur Verfügung. Heute ist das nicht mehr unbedingt "state of the art". Heute würde man die 
SDSF-Schnittstelle vorziehen und für Commandausführung ISFSLASH benutzen (siehe Beispiel 
dort). 

/* REXX 

   Übernimmt einen Befehl, führt ihn auf Console aus und reicht das 

   Ergebnis zurück. 

 

   Syntax im rufenden Programm: 

   interpret consinfo(command-+--------+---------------------->< 

                              '-\stem.-' 

 

   Nach dem Aufruf liegen die Variablen stem.0, stem.1 und folgende vor. 

   Wird kein Stem angegeben, liegt das Ergebnis im Stem CONSINFO. 

 

*/ 

parse source . rxTYPE RxName . 

 

   arg cmd "\" varname 

   if varname = "" then varname = "CONSINFO." 

   interpret varname "= ''" 

 

   "CONSPROF SOLDISP(NO) UNSOLDISP(NO)" 

 

   if rc <> 0 then do 

      say "CONSOLE NOT AVAILABLE" 

      return varname".0=0;" 

   end 

 

   /*************************************************************** 

   * Bilden eines eindeutigen Consolnamens innerhalb des Plexes   * 

   ***************************************************************/ 

   parse value time("L") with . "." suf 

   Str      = "ABCDEFGHIJKLMNOPQRSTUVWXYZ0123456789" 

   ConsName = substr(str,random(1,26),1)substr(str,random(1,36),1)suf 

   "CONSOLE ACTIVATE NAME("ConsName")" 

 

   ADDRESS CONSOLE cmd 

   /* wait n secs for solicited only */ 

   mrc = GETMSG(varname,"SOL",,,60) 

   "CONSOLE DEACT" 

   erg = varname"0 = "value(varname"0")";" 

   do §=1 to value(varname"0") 

      if rxTYPE = "COMMAND" then say value(varname"§") 

      erg = strip(erg varname !! § "= '"value(varname"§")"';") 

   end 

   if rxTYPE = "COMMAND" then exit 

RETURN erg 

1.29.19. Storage 

Die Storage-Funktion ist eines der "mystischen" Elemente in REXX. Die Funktion an sich ist nicht 
schwierig zu interpretieren: Mit Angabe einer Speicheradresse und einer Feldlänge wird direkt aus 
dem Arbeitsspeicher gelesen. Was die Funktion schwierig zu benutzen macht, ist das Wissen um 
den Aufbau des Speichers. Ich möchte an dieser Stelle nicht weiter gehen. Der Aufbau kann aus 
einschlägigen z/OS-Manuals nachgelesen werden. Dennoch möchte ich einige sehr hilfreiche Bei-
spielprogramme aufzeigen, welche mit Hilfe von Storage() nützliche Informationen beschaffen. 

An dieser Stelle nochmals herzlichen Dank an alle Member der Google-Newsgroup, von denen die 
meisten der Storage-Beispiele stammen. 

1.29.19.1. Jobname 

Kann interessant sein, wenn eine Verarbeitung ihr eigenes JobLog überprüfen muss und der Job-
name flexibel ist. 

/* REXX 

*/ 

CVT = STORAGE(10,4)            /* FLCCVT-PSA DATA AREA */ 

TCBP = STORAGE(D2X(C2D(CVT)),4)             /* CVTTCBP */ 

TCB = STORAGE(D2X(C2D(TCBP)+4),4) 

TIOT = STORAGE(D2X(C2D(TCB)+12),4)           /* TCBTIO */ 

return STRIP(STORAGE(D2X(C2D(TIOT)),8))    /* TIOCNJOB */ 

1.29.19.2. Programmname 

Sehr hilfreich, wenn eine Routine gegebenenfalls sowohl unter IKJEFT01 als auch unter IRXJCL 
läuft. Ist das Zweite der Fall, können nicht alle TSO-Kommandos und –Funktionen benutzt werden. 
In diesem Falle könnte gegebenenfalls auf USS-Services wie BPXWDYN ausgewichen werden. 

/* REXX 

Find program environment under MVS, Andy Wood 

*/ 

xp      = storage(0000021c,4) 

xp2     = ptr(xp,'b4'x) 

xp3     = d2x(c2d(xp2) + c2d('168'x)) 

envname = storage(xp3,8) 

return envname 

 

ptr: procedure 

   address = c2d(arg(1)) 

   offset  = c2d(arg(2)) 

return storage(d2x(address+offset),4) 

1.29.19.3. IPL-Informationen 

Liefert Informationen zum letzten IPL. 

/* REXX 

*/ 

tcb      = storage(21c,4) 

tiot     = storage(d2x(c2d(tcb)+12),4) 

jscb     = storage(d2x(c2d(tcb)+180),4) 

ssib     = storage(d2x(c2d(jscb)+316),4) 

jobname  = strip(storage(d2x(c2d(tiot)),8)) 

jobtype  = storage(d2x(c2d(ssib)+12),3) 

jobnum   = strip(storage(d2x(c2d(ssib)+15),5),l,0) 

stepname = strip(storage(d2x(c2d(tiot)+8),8),l,0) 

procstep = strip(storage(d2x(c2d(tiot)+16),8),l,0) 

 

CVT  = C2d(Storage(10,4))              /* point to CVT */ 

SMCA = Storage(D2x(CVT + 196),4)      /* point to SMCA */ 

SMCA = Bitand(SMCA,'7FFFFFFF'x) /* zero high order bit */ 

SMCA = C2d(SMCA)                 /* convert to decimal */ 

/*******************************************************/ 

/* 

IPL date is stored in packed decimal format - so to the  

date printable, it needs to be converted back to hex and 

the packed sign needs to be removed. 

*/ 

IPLTIME=C2d(Storage(D2x(SMCA + 336),4))  /*IPL Time-bin*/ 

IPLDATE=C2d(Storage(D2x(SMCA + 340),4)) /*Date-0CYYDDDF*/ 

IPLDATE= D2x(IPLDATE)           /* convert back to hex */ 

parse var IPLDATE . 2 IPLDATE 7 . 

iplday = date('w',ipldate,'j') 

parse value date('s',ipldate,'j') with 1 jjjj 5 mm 7 dd 

ipldate= dd"."mm"."jjjj 



Inside z/OS Seite 142 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

IPLTIME= IPLTIME / 100             /* remove hundreths */ 

HH = IPLTIME % 3600                        /* IPL hour */ 

MM = (IPLTIME - (3600 * HH)) % 60        /* IPL minute */ 

SS = (IPLTIME - (3600 * HH)- (60 * MM)) % 1 /* seconds */ 

HH = Right(HH,2,'0')              /* ensure 2 digit HH */ 

MM = Right(MM,2,'0')              /* ensure 2 digit MM */ 

SS = Right(SS,2,'0')              /* ensure 2 digit SS */ 

IPLTIME= HH':'MM':'SS             /* time format HH:MM */ 

say "Jobname......:" jobname"("jobtype""jobnum")" 

say "Step/Procstep:" stepname procstep 

say 

say "z/OS Level...:" left(mvsvar('sysmvs'),10) 

say "JES2.........:" sysvar('sysjes') 

say "System ID....:" mvsvar('sysname') 

say "IPL Volume...:" left(mvsvar('symdef','sysr1'),10) 

say "NJE Node.....:" left(sysvar('sysnode'),10) 

say "Proc.........:" sysvar('sysproc') 

say "IPL Date/Time:" iplday"," ipldate "at" ipltime 

exit 

 

1.29.19.4. Ist ein DD-Name zugewiesen? 

Manche Verarbeitungen werden gesteuert, in dem geprüft wird, ob ein bestimmtes DD-Statement 
zugewiesen ist. Dabei ist es unerheblich, ob der DD-Name auf eine bestimmte Ressource zeigt, 
oder DUMMY gesetzt ist. 

/* REXX * DDEXIST 

   Programmbeschreibung und Update-History am Programmende 

*/ 

   parse upper arg ddname . 

   parse source . RxTyp RxName . 

   dd_exist = 0 

   numeric digits 31  /* Need more digits for fullword 31bit addresses */ 

   tcb     = c2d(storage(21c,4)) 

   tiot    = c2d(storage(d2x(tcb+12),4)) 

   tiolen  = c2d(storage(d2x(tiot+24),1)) 

   tioddn  = storage(d2x(tiot+28),8) 

 

   do while(tiolen > 0) 

      if (left(tioddn,1) > " ") then, 

         if tioddn = ddname then do 

            dd_exist = 1 

            leave 

         end 

      tiot    = tiot + tiolen 

      tiolen  = c2d(storage(d2x(tiot+24),1)) 

      tioddn  = storage(d2x(tiot+28),8) 

   end 

 

   if RxTyp = "COMMAND" then do 

      say "DD Name '"ddname"' wurde geprüft, Return Code: "dd_exist 

   end 

exit dd_exist 

/* Programmbeschreibung --------------------------------------------- 

   Feststellen ob ein DD Name allokiert ist. 

   Eingabe: DDName 

   RC  0: DD Name nicht gefunden 

       1: DD Name ist allokiert 

 

   Gefunden im Internet, Autor unbekannt 

*/ 

1.29.19.5. ISPF-Screen-Informationen 

Ich habe diese Routine gebraucht, um auszuschließen, dass eine Verarbeitung im Split-Screen 
mehrfach gestartet wird. 

/* REXX - EXEC TO LIST ALL ACTIVE ISPF SCREENS 

*/ 

ADDRESS ISPEXEC "VGET ZSCREEN" 

TCBP=PTR(132+PTR(540)) /* CURRENT ISPTASK TCB       */ 

TCBP=PTR(TCBP+132)     /* ISPMAIN TCB VIA TCBOTC    */ 

TCBP=PTR(TCBP+136)     /* ISPTASK TCB VIA TCBLTC    */ 

DO WHILE TCBP <> 0 

   X=TLDID(TCBP) 

   INFO = '> TCB='D2X(TCBP)';  SCREEN NUMBER='SCRNR';  SCRNAME='SCRNAME 

   IF SCRNR=ZSCREEN THEN INFO=INFO "** ACTIVE SCREEN **" 

   SAY INFO 

   TCBP=PTR(128+TCBP)   /* FOLLOW NTC                */ 

END 

RETURN 

 

PTR: RETURN C2D(STORAGE(D2X(ARG(1)),4)) 

STG: RETURN STORAGE(D2X(ARG(1)),ARG(2)) 

 

TLDID: 

   R9=PTR(112+ARG(1)) 

   IF STG(R9+40,4)='ISPF' & STG(R9+24,1)='00'X THEN DO 

      R9= PTR(R9+24) 

      IF R9 <>0 & STG(R9+24,1)='00'X THEN DO 

         SCRNR  =STG(PTR(R9)+003,1) 

         SCRNAME=STG(PTR(R9)+852,8) 

         RETURN 0 

      END 

   END 

RETURN 'NONE' 

1.29.19.6. Wort an Cursorposition 

/* REXX  

*/ 

tcb    = PTR(540)               /* TCB (EXEC command)    PSATOLD  */ 

tcb    = PTR(tcb + 132)         /* TCB (ISPTASK)         TCBOTC   */ 

fsa    = PTR(tcb + 112)         /* first save area       TCBFSA   */ 

r1     = PTR(fsa + 24)          /* ISPTASK's R1                   */ 

tld    = PTR(r1)                /* TLD address                    */ 

tls    = PTR(tld + 096)         /* screen buffer         TLDTLSP  */ 

csr    = PTR(tld + 164)         /* relative cursor pos.  TLDCSR   */ 

csr    = csr + 1                /* Position (not offset)          */ 

scrw   = PTR(tld + 192)         /* screen width          TLDCLSWD */ 

scrl   = PTR(tld + 180) 

scradr = D2X(tls) 

screen = STORAGE(scradr,scrw * scrl) /* Content of current screen */ 

 

bpos = pos(" ",screen,csr)                                                       

tmp = left(screen,bpos-1)                                                        

parse value reverse(tmp) with tmp .                                              

tmp = reverse(tmp)                                                               

say ">"tmp"<"                                                                    

exit                                                                             

 

PTR: RETURN C2D(BITAND(STORAGE(D2X(ARG(1)),4),'7FFFFFFF'X)) 

1.29.20. Umwandlung von HTLM-Umlauten 

Das Beispiel unterstellt, dass in einer Datei, welche unter dem DD-Namen TELOUT zugewiesen ist, 
HTLM-Umlaute liegen, welche ersetzt werden sollen. Generell könnte das auch sehr gut mit einem 
Edit Makro gelöst werden (siehe Beispiel dort). 

/* REXX 

*/ 

parse source . . RxName . 

 

call set_vars 

"delstack" 

"execio * diskr TelOut (finis" 

 

do queued() 

   parse pull data 

   call change_html 

   queue data 

end 

queue 

"execio * diskw TelOut (finis" 

Exit 

 

changes: 

return (pos(AUml,data)+pos(OUml,data)+pos(UUml,data)+pos(SZ,DATA) > 0) 

 

change_html: 

   do while changes() 

      if pos(AUml,data) > 0 then do 

         parse var data _1 (Auml) _2 

         data = _1"Ä"_2 

      end 

      if pos(OUml,data) > 0 then do 

         parse var data _1 (Ouml) _2 

         data = _1"Ö"_2 

       end 

       if pos(UUml,data) > 0 then do 

         parse var data _1 (Uuml) _2 

         data = _1"Ü"_2 

      end 

      if pos(SZ,data) > 0 then do 

         parse var data _1 (SZ) _2 

         data = _1"ß"_2 

      end 

   end 

return 

 

set_vars: 

   AUml = "&AUML;" 

   OUml = "&OUML;" 

   UUml = "&UUML;" 

   SZ   = "&SZLIG;" 

Return 

1.29.21. TSO-User-Profile-Informationen 

Sehr häufig benötigt man Einstellungen aus dem TSO-Benutzerprofil. Um nicht an vielen Stellen 
immer wieder das gleiche Coding schreiben zu müssen, hier ein Beispiel für eine User Defined 
Function: 

/* REXX 

   Programmbeschreibung und Update History am Ende                               

*/ 

call msg "ON" 

parse source . RxType RxName . 

arg parm 

select 

   when abbrev("INTERCOM"  ,parm,3) then search = "INTERCOM" 

   when abbrev("MODE"      ,parm,3) then search = "MODE" 

   when abbrev("MSGID"     ,parm,3) then search = "MSGID" 

   when abbrev("PAUSE"     ,parm,3) then search = "PAUSE" 

   when abbrev("PREFIX"    ,parm,3) then search = "PREFIX" 

   when abbrev("PLANGUAGE" ,parm,3) then search = "PLANGUAGE" 

   when abbrev("PROMPT"    ,parm,3) then search = "PROMPT" 

   when abbrev("RECOVER"   ,parm,3) then search = "RECOVER" 

   when abbrev("SLANGUAGE" ,parm,3) then search = "SLANGUAGE" 

   when abbrev("VARSTORAGE",parm,3) then search = "VARSTORAGE" 

   when abbrev("WTPMSG"    ,parm,3) then search = "WTPMSG" 

   otherwise                             search = "PREFIX" 

end 

 

call get_prof_infos 

erg = value_of(search) 

if RxType <> "COMMAND" then return erg 

say search "ist" erg 

 

exit 

 

value_of: 

   found = pos(" "search,AllInfo) 

   if found > 0 then do 

      if substr(AllInfo,found + length(search)+1,1) = "(" then , 

         parse var AllInfo . =(found) . "(" erg ")" . 

      else erg = "ON" 

   end 

   else erg = "OFF" 

return erg 

 

get_prof_infos: 

   AllInfo = 

   call outtrap "PROF." 

   "PROFILE" 

   call outtrap "OFF" 

   do §=1 to prof.0 

      AllInfo = strip(AllInfo prof.§) 

   End 

Return 

 

/* Programmbeschreibung ********************************************** 

 

   Liefert Informationen aus dem TSO-Benutzerprofil. 

   Werden dort die Parameter mit [NO]parm geführt, wird ein ON oder 

   OFF zurückgegeben. 

   Sind die jeweiligen Werte in einer Klammer hinterlegt (wie PREFIX), 

   wird der Wert in Klammer oder OFF zurückgegeben. 

 

   Unterstützt werden folgende Parameter: 

 

   Parm------ Bedeutung----------------------------------------------- 

   CHAR       Nur der Vollständigkeit halber, wird bei Nutzung eines 

              Session Managers ignoriert. 

   INTERCOM   Bereit zum Informationsaustausch (z.B.: SEND-Messages) 

   LINE       Nur der Vollständigkeit halber, wird bei Nutzung eines 

              Session Managers ignoriert. 

   MODE       Nur der Vollständigkeit halber, wird nur in Verbindung 

              mit dem EDIT-Subcommand benutzt. 

   MSGID      Vor jeder Nachricht wird ein MSG-Identifier ausgegeben. 

   PAUSE      Nur der Vollständigkeit halber, wird unter TSO nicht 

              benutzt. 

   PREFIX     HLQ bei teilqualifizierten Dateinamen 

   PLANGUAGE  Primärsprache 

   PROMPT     Rückfragen bei teilweise falschen Befehlsparametern 

   RECOVER    Zeigt, ob im TSO-Edit die Recover-Option eingeschaltet ist 

   SLANGUAGE  Sekundär-Sprache 

   VARSTORAGE wird für Variablen der Bereich überhalb 16MB benutzt 

   WTPMSG     Entgegennahme von "Writge-to-Programmer-Messages" 

 

-- Update History ---------------------------------------------------- 

 

12.01.2015 Wittemann, Erstellung 

*/ 

1.29.22. 64Bit Adressierung, oder wie viel sind 16 Exabyte 

Es ist immer wieder erstaunlich, wie selbstverständlich über 64Bit-Adressierung gesprochen wird. 
Aber wer hat die richtige Vorstellungskraft, was sich aus 64Bit überhaupt ergibt? Es sind 16Exabyte. 
Und wie viel ist das nun wieder? Das nachfolgende Beispiel versucht dies zu veranschaulichen: 

/* REXX 

   Was ist ist 64 BIT? Berechnung von 16EB, wenn 1 Byte = 1 Millisekunde 

*/ 

arg Type 

if wordpos(Type,"D B") = 0 then Type ="B" 

 

numeric digits 30 

ms          = 0.001                              /* 1 ms = 1 Byte */ 

base = word("1000 1024",pos(Type,"DB")) 

Art.B = "Binary  Prefix (1KB = 1024 Byte)" 

Art.D = "Decimal Prefix (1KB = 1000 Byte)" 

 

say;say left("--------< Wieviel ist 64 Bit ? >",linesize()-1,"-");say 

say;say "Gerechnet wird mit" Art.Type 

say;say "Basis: 1Byte = 1ms" 

say     "       1KB   = 1 Sekunde" 

say     "       1MB   =" output(base**1) 

say     "       1GB   =" output(base**2) 

say     "       1TB   =" output(base**3) 

say     "       1PB   =" output(base**4) 

say     "       1EB   =" output(base**5) 

say;say "64bit=16EB   =" output(base**5*16);say 

say;say left("--------< Kaum zu glauben, was? >",linesize()-1,"-") 

exit 

 

output: 

   erg =  

   Jahre=arg(1)/(60*60*24*365);parse var jahre jahre"."dec /* Jahre  */ 

   Tage =365 * value("."dec)  ;parse var tage  tage"."dec  /* Tage   */ 

   std  = 24 * value("."dec)  ;parse var std   std"."dec   /* Std.   */ 

   min  = 60 * value("."dec)  ;parse var min   min"."dec   /* Min.   */ 

   sec  = 60 * value("."dec)  ;parse var sec   sec"." .    /* Sek.   */ 

   if sec   > 0 then erg = sec  "Sekunden, " erg 

   if min   > 0 then erg = min  "Minuten, "  erg 

   if std   > 0 then erg = std  "Stunden, "  erg 

   if tage  > 0 then erg = tage "Tage, "     erg 

   if jahre > 0 then erg = §format(jahre,"D",,0) "Jahre," erg 

   erg = space(strip(strip(erg),"T",","),1) 

return erg 

1.29.23. Directory-Informationen mit REXX-Mitteln 

Üblicherweise ermittelt man Directory-Informationen von PO (PO/E-1 und 2) über die LM-Services. 
Aber was tun, wenn ISPF nicht verfügbar ist? We do it in REXX! (Doug Nadel war der Nachfolger 
von Mike Cawlishaw, dem REXX-Erfinder. Ist leider auch schon verrentet) 

/* Rexx * routine to read a pds directory 

*/ 

Parse Upper Arg dsname . 

If listdsi(dsname) <> 0 Then Do 

    Say 'Data set 'dsname' not allocated' 

    Exit 

End 

If substr(sysdsorg,1,2) <> "PO" Then do 

   Say 'Data set 'dsname' is not a pds' 

   Exit 

End 

'ALLOC F(AREAD) DS('dsname') SHR REUSE DSORG(PS) LRECL(256) RECFM(F B)' 

'EXECIO * DISKR AREAD ( FINIS STEM DIRS.' 

'FREE F(AREAD)' 

Do blocknum = 1 to dirs.0 

   block=dirs.blocknum 

   Parse Var dirs.blocknum bl 3 block 

   block=substr(block,1,c2d(bl)-2) 

   Do While block<>'' 

      Parse Var block name 9 ttr 12 c 13 block 

      c=c2d(bitand(c,'1f'x)) 

      If name='FFFFFFFFFFFFFFFF'x Then Leave blocknum 

      stats='' 

      If c=15 & sysrecfm ^='U' Then stats=decodestats(substr(block,1,30)) 

      Say name stats 

      block=delstr(block,1,c*2) 

   End 

End 

Exit 

decodestats:  Procedure  /* decode the directory */ 

   direntry=Arg(1) 

   Parse Var direntry, 

      vv 2 mm 3 flags 4 ss 5 crecc 6 crdate 9 modcc 10 moddate 13 hh, 

      14 tm 15 lines 17 init 19 mod 21 userid 28 . 

   Parse Value c2x(crdate)  with cyy 3 cjjj 6 . 

   Parse Value c2x(moddate) with myy 3 mjjj 6 . 

   out=leftd(vv)'.'leftd(mm)                       /* Vv.Mm         */ 

   out=out 19+c2x(crecc)cyy'.'cjjj                 /* Create date   */ 

   out=out 19+c2x(modcc)myy'.'mjjj                 /* Mod date      */ 

   out=out leftx(hh)':'leftx(tm)':'leftx(ss)       /* mod time      */ 

   out=out right(c2d(lines),5)                     /* Lines         */ 

   out=out right(c2d(init),5)                      /* Initial lines */ 

   out=out right(c2d(mod),5)                       /* Mod lines     */ 

   out=out userid                                  /* User id       */ 

   If bitand(flags,'80'x)='80'x Then out=out '(SCLM)'/* Saved by SCLM */ 

Return out 

leftd: return right(c2d(arg(1)),2,'0')                                           

leftx: return right(c2x(arg(1)),2,'0') 

1.29.24. String-Checker 

Needle-Haystack-Funktion vergleichbar mit POS(). Im Unterschied dazu kann NEEDLE aber 
Wildcards enthalten. 

/* REXX StringCk 

   does string contain mask?  

*/ 

String = 'The quick brown fox  jumps over the lazy dog' 

ruler  = '1...5....0....5....0....5....0....5....0...4' 

Mask.1 = '%%E%%%%%%%BROWN' 

Mask.2 = 'QUICK*F%X' 

Mask.3 = '*t ' 

Mask.4 = 'THE%Q%*%**BR%%%' 

Mask.5 = '  E*BROWN' 

Mask.6 = 'THE*BRXWN' 

Mask.7 = 'THE*BROWN*F%%' 

Mask.8 = '*FOX*' 

Mask.9 = '*F%X*' 

Mask.10= '*F%*%MP*G' 

Mask.11= 'T*E LAZY DOG' 

Mask.12= '*E LAZY DOG' 

Mask.13= '*  JU' 

Mask.14= 'O%E' 

Mask.15= '%*V*D' 

Mask.16= 'G*' 

Mask.17= '*G%' 

say string 

say ruler 

Do m = 1  to 17 

   If StringCK(string,mask.m,1) then Status = 'Found     ', 

      right(BgnStr,3) Right(EndStr,3), 

      substr(string,BgnStr,EndStr+1-BgnStr) 

   else Status = 'NOT Found' z 

   say left(Mask.m,25) Status 

End 

exit 0 

 

StringCK: 

   Arg stringx,maskx,strt 

   /* Adjust mask: 

      force * at begining of mask 

      if there is an * in a set of wild characters it should be the last 

      char in the set ie %%*%**%% s/b %%%%%* 

      only one * is needed within each set of wild characters 

      an ending * is not needed 

*/ 

   maskx = '*'maskx 

   Do while pos('*%',maskx) > 0 

      maskx = overlay('%*',maskx,pos('*%',maskx)) 

   End 

   Do while pos('**',maskx) > 0 

      maskx = delstr(maskx,pos('*',maskx),1) 

   End 

   maskx = strip(maskx,'T','*') 

   LenMask = length(maskx) 

   If datatype(strt) <> "NUM" then strt = 1 

   If strt < 1 ! strt > length(stringx) then strt = 1 

   Nxtj = strt-1   /* The next position (-1) of the string to check */ 

   Nxti = 1        /* The next position of the mask to match        */ 

   BgnStr = 0 

   EndStr = 0 

   Do z = 1 to length(stringx) /* z = number of iterations */ 

      i = Nxti 

      j = Nxtj 

      Do while j < length(stringx) 

         j = j+1 

         Select 

            when substr(maskx,i,1) = '*' then Do 

               Savei = i      /* save where we are in mask and string */ 

               Savej = j 

               SaveBgnStr = BgnStr 

               /* look for mask/string match after an *,              */ 

               /*   if found set pointers to continue looking         */ 

               If substr(maskx,i+1,1) == substr(stringx,j,1) then Do 

                  i=i+1 

                  j=j-1 

               End 

            End 

            when substr(maskx,i,1) = '%' !, 

               substr(maskx,i,1) == substr(stringx,j,1) then Do 

               If BgnStr = 0 then BgnStr = j 

               EndStr = j 

               i=i+1 

               If i > LenMask then Return 1 

            End 

            /* mismatch, go back to last * and look again */ 

            otherwise Do 

               BgnStr = SaveBgnStr 

               Nxti = Savei 

               Nxtj = Savej 

               iterate z 

            End 

         End 

      End 

   Return 0 

   End 

Return 0 

1.29.25. String Counter 

Prüft, wie oft eine Zeichenfolge in einem String vorkommt. 

/* REXX 

   Programmbeschreibung und Update History am Ende 

*/ 

parse source . RxTyp RxName . 

parse arg needle, haystack, spos 

if ^Datatype(spos,"W") then SPos = 1 

haystack = substr(haystack,spos) 

do §=0 while pos(needle,haystack) > 0 

   parse var haystack (needle) haystack 

end 

if RxTyp = "COMMAND" & sourceline() > 0 then say needle § "mal gefunden" 

return § 

 

/* Programmbeschreibung ********************************************** 

 

   REXX to return number of character 'needle' in 'haystack' 

 

   Notes 

   ----- 

   Smart use of parse, written by Dallas Legan as he sat with Mike 

   Cowlishaw, http://www.scoug.com/os24u/2002/scoug202.changestr.html 

 

   If you inline it, make it countstr: procedure 

 

   History 

   ------- 

   Possibly by Dallas Legan & possibly with help from Mike Cowlishaw 

   20 March 2015 UASNH-tidy, comments 

 

-- Update History ---------------------------------------------------- 

 

   06.07.2018 Wittemann 

              ° Umstellung von Rekursion auf LOOP, da es bei VIELEN! 

                Rekursionen zu Problemen in den Kontrollblöcken kommt. 

*/ 

  

http://www.scoug.com/os24u/2002/scoug202.changestr.html

1.29.26. String Update 

Funktioniert wie ein "CHANGE ALL"-Befehl im Editor 

/* REXX 

*/ 

parse source . RxTyp RxName . 

if RxTyp <> "FUNCTION" then , 

   call syntax RxName "kann nur als Function benutzt werden." 

/* 

new string to be used ---------+ +------? limit of # changes (times). 

original string (haystack) --+ | |        [default: one billion] 

old string to be replaced -+ | | | +----? begin at this occurrence # 

(O/H/N can be NULL)        | | | | | +--? start position (default=1) 

                           | | | | | | 

                           | | | | | |*/ 

parse arg                  o,h,n,t,b,p, ,$,F /* T, B, P = optional  */ 

t=word(t 999999999, 1)                  /* maybe use default        */ 

b=word(b 1        , 1)                  /* maybe use default        */ 

p=word(p 1        , 1)                  /* maybe use default        */ 

if arg() < 3         then call syntax   "Zu wenig Parameter empfangen" 

if arg() > 6         then call syntax   "Zu viele Parameter empfangen" 

if ^datatype(t, 'W') then call syntax   "4. Parameter kein INT-Wert" 

if ^datatype(b, 'W') then call syntax   "5. Parameter kein INT-Wert" 

if ^datatype(p, 'W') then call syntax   "6. Parameter kein INT-Wert" 

if t<0               then call syntax   "4. Parameter ist Negativwert" 

if b<1               then call syntax   "5. Parameter muss > 0 sein" 

if p<1               then call syntax   "6. Parameter muss < 0 sein" 

L=length(o)                             /* length of OLD string     */ 

if L==0  &  t^=0     then return n !! h /* changing a [null] char?  */ 

                                        /* check for position       */ 

if p^=1  then do                        /* Pí=1?  Then ajust F & H. */ 

   f=left(h, min(p-1, length(h)))       /* keep first part intact.  */ 

   h=substr(h, p)                       /* only use this part of H. */ 

end                                     /* now, proceed as usual.   */ 

#=0                                     /* # of changed occurrences */ 

do j=1 while # < t                      /* keep changing, T times.  */ 

   parse var  h  y  (o)  _  +(L)  h     /* parse the haystack ...   */ 

   if _==''  then return f !! $ !! y    /* no more left, return.    */ 

   $=$ !! y                             /* append the residual txt. */ 

                                        /* check if too soon.       */ 

   if j<b    then $=$ !! o              /* append OLD if too soon.  */ 

   else do                              /* met the occurrence test. */ 

     $=$ !! n                           /* append the NEW  string.  */ 

      #=#+1                             /* bump occurrence number.  */ 

   end                                  /* append new string.       */ 

end   /*j*/                             /* Note:  most REXX         */ 

return  f !! $ !! h                     /* support three options.   */ 

 

syntax: say arg(1);exit "" 

 

/* Update-History ----------------------------------------------------- 

 

   05.07.2018 Wittemann 

              °Aufgehübscht und angepasst 

              °Gefunden im Internet auf http://www.scoug.com/ 

*/ 

1.29.27. Dateien zu einem DD-Namen 

/* REXX * Ermitelt DSNs zum DD-Namen und gibt sie als Wortkette zurück 

*/ 

parse source . rxTYP RxName . 

arg dd_name . 

call outtrap "lista." 

"lista st" 

call outtrap "OFF" 

do §=1 to lista.0 while wordpos(dd_name,lista.§) = 0 

end 

dalist = 

if § <= lista.0 then do 

   §§=§-1 

   dalist = dalist strip(lista.§§) 

   do §=§+2 to lista.0 by 2 

      §§=§-1 

      if substr(lista.§,3,8) <> " " then leave 

      dalist = dalist strip(lista.§§) 

   end 

end 

if rxTYP <> "COMMAND" then return strip(dalist) 

say "Dateien zu" dd_name;SAY 

do i=1 to words(dalist) 

   say word(dalist,i) 

end 

exit 

1.29.28. SubMixed 

Häufig gibt es Probleme, wenn JCL über den TSO-Befehl SUBMIT zur Ausführung gebracht werden 
soll. Zum einen wird alles auf Uppercase umgesetzt, zum anderen sind keine Sätze größer 80 Byte 
möglich. Dem kann abgeholfen werden: 

/* REXX 

   Demo SUBMIT über EXECIO 

   erlaubt auch MixedCase und SL > 80 

*/ 

§=0 

§=§+1;jcl.§="//SP1A JOB (RZRZ),NOTIFY=SP1,MSGCLASS=H" 

§=§+1;jcl.§="// EXEC PGM=IEFBR14" 

§=§+1;jcl.§="//* Das ist Kommentar in Mixed Case" 

"alloc dd(jcl) sysout(A) writer(intrdr)" 

"execio * diskw jcl (stem jcl. finis" 

"free dd(jcl)" 

  

http://www.scoug.com/


Inside z/OS Seite 154 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

1.29.29. Wie lange dauert WAS? 

Oft gibt es viele Möglichkeiten zu einem Ergebnis zu kommen, aber nicht immer ist die erste Idee 
auch die beste. Die folgenden Routinen zeigen für gleiche Verarbeitung unterschiedliche Lösungen 
und melden, wie lang die Ausführung dauert. 

1.29.29.1. EXECIO 

/* REXX 

   Demo für EXECIO - Einzelsätze immer langsamer 

   Die Eingabedatei sollte mindestens 1000 Datensätze enthalten 

*/ 

"alloc dd(tmp) da(your_input_dataset) shr reuse" 

call time "R" 

do forever 

   "execio 1 diskr tmp (stem demo." 

   if rc > 0 then leave 

end 

"execio 0 diskr tmp (finis" 

say "Einzele Records:" time("E") 

 

call time "R" 

"execio * diskr tmp (stem demo. finis" 

say "Lesen mit *    :" time("E") 

exit 

1.29.29.2. Zerlegen Wortkette 

/* REXX 

   Beispiele für Verbraten von CPU 

*/ 

Text = "auf  dem Baum     sitzen 20   Flaschen  Bier    " 

Loop = 1e3 

call time "R" 

do Loop 

   do i=1 to words(Text) 

      var = word(Text,i) 

   end 

end 

say "Zerlegen mit WORD():" time("E");call time "R" 

 

do Loop 

   do forever 

      parse var Text var Text 

      if Text = "" then leave 

   end 

end 

say "Zerlegen mit PARSE :" time("E");call time "R" 

exit 

1.29.29.3. Stem oder Stack? 

/* REXX 

   Zeitverhalten Stem./STACK 

*/ 

"alloc dd(data) da('sp1.edu.data(adrsort)') shr reuse" 

call time "R" 

do forever 

   drop satz. 

   "execio 5 diskr data (Stem satz." 

   do i=1 to satz.0 

      a=satz.i 

   end 

   if rc > 0 then leave 

end 

"execio 0 diskr data (finis" 

say "EXECIO je 5 Sätze via Stem " time("E");call time "R";say 

 

do until RC > 0 

   "execio 1 diskr data (Stem satz." 

   a=satz.1 

end 

"execio 0 diskr data (finis" 

say "EXECIO je 1 Satz  via Stem " time("E");call time "R";say 

 

"execio * diskr data (stem Satz." 

do i=1 to satz.0 

    a=satz.i 

end 

say "EXECIO *    Sätze via Stem " time("E");call time "R";say 

 

"execio * diskr data (finis" 

do queued() 

    parse pull satz 

end 

say "EXECIO *    Sätze via Stack" time("E");call time "R" 

1.29.29.4. Parse versus Substr 

/* REXX  

*/ 

loops = 20 

do loops 

   say 

   call time "R" 

   do loops 

      t   = substr(date(),1,2) 

      m   = substr(date(),4,3) 

      j   = substr(date(),8,4) 

      std = substr(time(),1,2) 

      min = substr(time(),4,2) 

      sec = substr(time(),7,2) 

   end 

   say "Substring:" time("E") 

   call time "R" 

   do loops 

      parse value date() time() with t m j std ":" min ":" sec 

   end 

   say "Parse    :" time("E") 

end 

1.29.29.5. Interpret versus Value 

Sehr häufig wird INTERPRET benutzt, wo es nicht nötig ist. In manchen Unternehmen ist die Nut-
zung von INTERPRET sogar untersagt. 

/* REXX  

*/ 

loop1 = 1 

loop2 = 1e4 

do loop1 

   say 

   call time "R" 

   do loop1 

      call loop 

   end 

   say "Loop          :" time("E") 

   call time "R" 

   do loop1 

      call loop_interpret 

   end 

   say "Loop Interpret:" time("E") 

   call time "R" 

   do loop1 

      call loop_value 

   end 

   say "Loop Value    :" time("E") 

end 

exit 

 

loop: 

   do §=1 to loop2 

      var.§ = "gaga" 

   end 

return 

 

loop_interpret: 

   do §=1 to loop2 

      interpret "var."§ "= gaga" 

   end 

return 

 

loop_value: 

   do §=1 to loop2 

      call value "var."§,"gaga" 

   end 

return 

1.29.29.6. Single UPPER versus multiple Upper 

Was man oft sehen kann ist, dass "der Schönheit wegen", mehrfach einzelne Befehle untereinander 
kodiert stehen, wo EIN Befehl die gesamte Arbeit erledigen kann. Im Fall des TSO-FREE-Befehls 
beispielsweise do, dass 4 FREE-Befehle, die jeweils einen DD-Namen freigeben, beinahe viermal 
so lange dauern als ein FREE-Befehl, der vier DD-Namen freigibt. Das folgende Beispiel zeigt die 
Unterschiede bei der UPPER-Instruktion: 

/* REXX 

*/ 

Loops = 1e4 

do 10 

   say 

   call time "R" 

   do loops 

      call mult_upper 

   end 

   say "x UPPER:" time("E") 

   call time "R" 

   do loops 

      call one_upper 

   end 

   say "1 UPPER:" time("E") 

end 

exit 

 

mult_upper: 

   UPPER HA ; UPPER OA 

   UPPER HB ; UPPER OB 

   UPPER HC ; UPPER OC 

   UPPER HD ; UPPER OD 

   UPPER HE ; UPPER OE 

   UPPER HF ; UPPER OF 

   UPPER HG ; UPPER OG 

   UPPER HH ; UPPER OH 

   UPPER HI ; UPPER OI 

   UPPER HJ ; UPPER OJ 

   UPPER HK ; UPPER OK 

   UPPER HL ; UPPER OL 

   UPPER HM ; UPPER OM 

   UPPER HN ; UPPER ON 

   UPPER HO ; UPPER OO 

   UPPER HP ; UPPER OP 

   UPPER HQ ; UPPER OQ 

Return 

 

one_upper: 

   UPPER HA HB HC HD HE HF HG HH HI HJ HK HL HM HN HO HP HQ , 

         OA OB OC OD OE OF OG OH OI OJ OK OL OM ON OO OP OQ 

Return 

1.29.29.7. Parse versus Translate versus Upper 

Das Umformen einen Variableninhalts auf UpperCase dauert weniger lange, wenn der Inhalt bereits 
UpperCase ist. Wir betrachten hier ein paar unterschiedliche Varianten. 

/* REXX 

   PARSE vs. TRANSLATE() vs. UPPER 

   Generell: ein Function-Call ist immer zeitaufwändiger 

*/ 

loops = 1e5 

var = "plumpaquatsch" 

call time "R" 

do loops 

   b = translate(var) 

end 

say "Translate:" time("E") 

call time "R" 

do loops 

   parse upper var var var 

end 

say "Parse....:" time("E") 

call time "R" 

do loops 

   upper var 

end 

say "Upper....:" time("E") 

say "Jetzt ist var IMMER lowercase" 

call time "R" 

do loops 

   var = "plumpaquatsch" 

   b = translate(var) 

end 

say "Translate:" time("E") 

call time "R" 

do loops 

   var = "plumpaquatsch" 

   parse upper var var var 

end 

say "Parse....:" time("E") 

call time "R" 

do loops 

   var = "plumpaquatsch" 

   upper var 

end 

say "Upper....:" time("E") 

exit 

1.29.29.8. Umformatieren Datum (selbstgebasteltes versus translate()) 

Um ein Datum in die bei uns übliche Schreibweise (tt.mm.jjjj) zu bringen, gibt es viele Möglichkeiten. 
Das folgende Beispiel zeigt das Zerlegen per SubStr() und das anschließende wieder zusammen-
bauen im Gegensatz zur Translate-Funktion: 

/* REXX  

*/ 

loops = 1e4 

say "Zeitvergleich Datum umbauen" 

do 1 

   say 

   call time "R" 

   do loops 

      t     = substr(date("S"),7,2) 

      m     = substr(date("S"),5,2) 

      j     = substr(date("S"),1,4) 

      datum = t"."m"."j 

   end 

   say "Selbst gebastelt:" time("E") 

   call time "R" 

   do loops 

      datum = translate("Tt.Mm.CcJj",date("S"),"CcJjMmTt") 

   end 

   say "Mit Translate() :" time("E") 

end 

exit 

1.29.29.9. Command-Outputs umleiten 

HELP ist eine praktische Art, sich online Hilfe für Befehle anzeigen zu lassen, Schade, dass es so 
unkomfortabel ist. Das Beispiel zeigt eine kleine Routine, welche Bildschirmausgaben einfängt und 
das Resultat im View anzeigt. 

/* REXX 

*/ 

parse source . . RxName . 

 

arg cmd 

address isredit "macro (parm)" 

if rc = 0 then cmd = translate(parm) 

arg cmd "/" how2show 

cmd = strip(cmd) 

how2show = strip(how2show) 

if wordpos(word(cmd,1),"B E V") > 0 then parse var cmd how2show cmd 

call outtrap info. 

interpret cmd 

call outtrap "OFF" 

do i=1 to info.0 

   if length(info.i) > lrecl then lrecl = length(info.i) 

end 

call msg "OFF" 

parms = "lrecl("lrecl") dsorg(PS) recfm(F) space(15) tracks" , 

        "rtddn(ddn) rtdsn(dsn)" 

call bpxwdyn "alloc" parms 

"execio * diskw" ddn "(stem info. finis" 

call edivibro ddn "/"How2Show 

call bpxwdyn "free dd("ddn")" 

exit 

 

/* Programmbeschreibung *********************************************** 

 

   TRAP Output des ARGs und Anzeige über Tempfile im VIEW 

 

   Anstatt ZTEMPF wird eigenes Dataset allokiert, dessen LRECL 

   der längsten erzeugten Ausgabevariablen entspricht. 

 

   Wird als letzter Parameter ein /B, /E oder /V geliefert, erfolgt 

   Anzeige im Browse, Edit oder View(Default). 

 

-- Update-History ----------------------------------------------------- 

 

   18.12.2013 Wittemann, Erstellung 

 

*/ 

1.30. REXX Fehlermeldungen 

Die nachfolgenden Fehlernummern werden vom Interpreter bei Auftreten eines Syntaxfehlers aus-
gegeben. Der Interpreter meldet die Fehlernummer mit dem Vorspann IRXxxxxI und einer kurzen 
Erläuterung. Wurde in einem Programm SIGNAL ON SYNTAX kodiert, liegt die jeweilige Fehlernum-
mer in der Variablen RC. 

IRX0003I Program is unreadable 

Erklärung: Das Programm konnte nicht gelesen werden. Die Ursache hierfür ist häufig der 
Aufruf von IRXEXEC mit Übergabe eines Programmes, das wegen falschen Formates nicht 
gelesen werden kann. 

Systemreaktion: Programm-Abbruch 

Bediener-Aktion: Prüfen Sie das Format des Programmes oder benachrichtigen Sie die Sys-
temprogrammierung. 

IRX0004I Program interrupted 

Erklärung: Die Programmausführung wurde unterbrochen. Im Allgemeinen ist die Ursache 
hierfür die Eingabe HI auf die PA-1-Taste hin, oder der Befehl EXECUTIL HI innerhalb des 
Programmes. Liegt die Ursache in einem anderen Fehler, wird eine entsprechende Fehler-
meldung ausgegeben. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Sofern der Abbruch nicht gewollt war, beheben Sie Ihr Problem aufgrund 
der ausgegebenen Fehlermeldung. 

IRX0005I Machine storage exhaust 

Erklärung: Speicherüberlauf. In aller Regel wird der Fehler durch eine Fehlerschleife eines 
Programmes erzeugt, bei der Speichervariablen in endloser Zahl erzeugt werden und der In-
terpreter nicht genügend Speicher zur Verfügung gestellt bekommt. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Prüfen Sie die Ursache für den Loop (TRACE).  

IRX0006I Unmatched  "/*" or quote 

Erklärung: Der Interpreter hat Dateiende (oder Datenende bei einer INTERPRET-Instruktion) 
erreicht, ohne auf ein Kommentar- oder Literalbegrenzungszeichen gestoßen zu sein. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm, indem Sie das fehlende Begrenzungszei-
chen hinzufügen. Fällt es Ihnen schwer, die Ursache zu finden, kodieren Sie das Statement 
TRACE SCAN an den Programmanfang und starten Sie Ihr Programm erneut.  

IRX0007I WHEN or OTHERWISE expected 

Erklärung: Der Interpreter findet in einem SELECT-Gebilde keine WHEN- oder OTERWISE-
Klausel. Häufige Ursache ist das Fehlen eines END-Statements (siehe Beispiel). 

Falsch Richtig 

SELECT SELECT 
 WHEN ? THEN  WHEN ? THEN DO 
 SAY ...   SAY ... 
 EXIT   EXIT 
 OTHERWISE NOP  END 
END  OTHERWISE NOP 
:  END 
Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0008I Unexpected THEN or ELSE 

Erklärung: Der Interpreter fand eine THEN- oder ELSE-Klausel, für die es keine Zuordnung 
zu einem IF-Befehl gibt. Häufige Ursache ist das Fehlen eines END-Statements (siehe Bei-
spiel). 

Falsch Richtig 

IF ? THEN IF ? THEN DO 
 SAY ...   SAY ... 
 EXIT   EXIT 
ELSE SAY ... END 
:  ELSE SAY ... 
Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0009I Unexpected WHEN or OTHERWISE 

Erklärung: WHEN oder OTHERWISE außerhalb eines SELECT gefunden. Häufige Ursache 
ist der Versuch, durch einen Sprung mit SIGNAL das SELECT-Gebilde zu verlassen (in die-
sem Fall wird SELECT als unmittelbar beendet betrachtet). 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0010I Unexpected or unmatched END 

Erklärung: Mehr END-Statements, als DO- oder SELECT-Anweisungen vorhanden, oder die 
ENDs stehen in einer Reihenfolge, die keine Zugehörigkeit zu DO- oder SELECT erkennen 
lassen.  

Meist zweierlei Gründe: Auf THEN oder ELSE folgt unmittelbar END. Mit SIGNAL wurde in 
eine Schleife gesprungen. Wird das Schleifenende gelesen, kann keine Beziehung zum 
Schleifenkopf hergestellt werden (SIGNAL ist kein GOTO-Befehl). 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. TRACE SCAN hilft. 

IRX0011I Control stack full 

Erklärung: Diese Meldung erscheint, wenn die mögliche Verschachtelungstiefe von 250 für 
DO-END oder IF-THEN-ELSE überschritten wurde, oder ein LOOP durch eine Anweisung wie 

befehl="INTERPRET befehl" 

INTERPRET befehl 

kodiert wird. Hier erzeugt der INTERPRET-Befehl die Schleife, die zur Meldung führt, wenn 
sie mehr als 250 mal durchlaufen wird. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0012I Clause > 500 characters 

Erklärung: Befehllslänge überschreitet 500 Zeichen (Ohne Kommentare -  mehrfache Blanks 
werden wie eines betrachtet). 

Meist ein fehlendes Literal-Begrenzungszeichen. In diesem Fall erscheint die Meldung gleich 
zu Beginn der Programmverarbeitun (wird durch drei führende Plus(+) markiert). 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0013I Invalid character in data 

Erklärung: Nicht erlaubtes Zeichen außerhalb eines Literals. Häufig sind dies im deutsch-
sprachigen Raum die Umlaute.  

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0014I Incomplete DO/SELECT/IF 

Erklärung: Der Interpreter hat Programmende erreicht, ohne dass für jedes DO oder SELECT 
ein END, oder für IF ein zugehöriges THEN erkannt wurde. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0015I Invalid hex constant 

Erklärung: Hex-Konstanten dürfen weder führende noch nachlaufende Blanks enthalten. 
Blanks dürfen ausschließlich zur Trennung an der Bytegrenze innerhalb der Konstanten ko-
diert werden. 

'14'x 

'C8 E3 C7 D6'x 

'2df4'x 

Häufige Fehler sind das Kodieren des Buchstaben O statt der Zahl 0, oder eine Variable X 
wird unmittelbar im Anschluss an ein Literal kodiert (das Literal wird so zu einer HEX-Konstan-
ten). 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0016I Label not found 

Erklärung: Das Ziel einer Sprunganweisung SIGNAL kann nicht gefunden werden. Häufig 
liegt es an Schreibfehlern, oder das Label wurde vergessen (SIGNAL ON SYNTAX reklamiert 
ein fehlendes Label SYNTAX erst, wenn Syntaxfehler auftritt). 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0017I Unexpected PROCEDURE 

Erklärung: Die Anweisung PROCEDURE wurde an falscher Stelle entdeckt. Der häufigste 
Fehler hierfür: PROCEDURE ist nicht die erste Instruktion, die durch CALL oder Funktions-
aufruf angesteuert wird. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0018I THEN expected 

Erklärung: Alle IF und WHEN-Klauseln verlangen zwingend den Operanden THEN. Entweder 
fehlt die Angabe, oder ein anderes Wort wurde vor THEN kodiert. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0019I String or Symbol expected 

Erklärung: Die Instruktionen CALL, SIGNAL, SIGNAL ON oder SIGNAL OFF erwarten zwin-
gend ein Symbol oder Literal im Anschluss. Entweder wurde es vergessen oder es enthält ein 
an dieser Stelle nicht erlaubtes Sonderzeichen wie "(". 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0020I Symbol expected 

Erklärung: Befehle wie DROP, PARSE oder UPPER erwarten zwingend ein oder mehrere 
Symbole hinter dem Befehlswort. Das Symbol wurde nicht gefunden. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0021I Invalid data on end of clause 

Erklärung: Eine Instruktion wie NOP oder SELECT wird von etwas anderem als einem Kom-
mentar gefolgt. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0022I Invalid character string 

Erklärung: Eine unzulässige Zeichenfolge wurde entdeckt (siehe REXX-REFERENCE-Ma-
nual, Befehl OPTIONS - exotisch) 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0023I Invalid SCBS/DBCS mixed string 

Erklärung: Siehe vorhergehende Meldung. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0024I Invalid TRACE request 

Erklärung: Eine ungültige TRACE-Option wurde kodiert. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0025I Invalid sub-keyword found 

Erklärung: Der Befehlsprozessor erwartet hinter bestimmten Befehlen ganz bestimmte Sub-
Parameter. So ist beispielsweise hinter NUMERIC einer der Werte DIGITS, FUZZ oder FORM 
zu kodieren. Erkennt der Befehlsprozessor irgend etwas anderes, erscheint diese Meldung. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0026I Invalid whole number 

Erklärung: Kein numeric integer Wert, wo einer erwartet, oder der Maximalwert von 999 999 
999 wurde überschritten. Dies kann beispielsweise in der Anweisung NUMERIC oder auf der 
rechten Seite der Potenzierung (**) vorkommen. 

Ebenso möglich, wenn das Programm als Command aufgerufen wurde und in der Anweisung 
EXIT oder RETURN eine Variable zurückgereicht wird. Entweder ist die Variable falsch gefüllt, 
oder es handelt sich um einen Schreibfehler wie EXIT RT statt EXIT RC. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0027I Invalid DO syntax 

Erklärung: Der Befehlsprozessor entdeckte einen Syntaxfehler in einem Schleifenkopf. Dies 
kann die doppelte Kodierung von BY oder TO sein, oder BY oder TO waren angegeben, aber 
eine Steuervariable war nicht definiert. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0028I Invalid LEAVE or ITERATE 

Erklärung: LEAVE oder ITERATE wurden als falsch erkannt, weil sie entweder außerhalb 
einer Schleife stehen oder der Name hinter der Anweisung nicht mit einer Steuervariablen ei-
nes Schleifenkopfes übereinstimmt. 

Bedenken Sie, dass mit einer Instruktion LEAVE in einer Subroutine keine Schleife im Haupt-
programm verlassen werden kann. Auch wenn Sie SIGNAL zum GOTO vergewaltigen, kann 
es zu dieser Meldung kommen. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0029I Environment name to long 

Erklärung: Der Name einer Systemumgebung ist länger als 8 Byte. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihre ADDRESS-Instruktion. 

IRX0030I Name or string > 250 Byte 

Erklärung: Ein Literal oder ein Variablenname größer 250 Byte wurde entdeckt. Häufigste 
Ursache ist das Fehlen eines Begrenzungszeichens für ein Literal. Müssen Variablen mit ei-
nem Literal > 250 Byte gefüllt werden, muss dies in mehreren Schritten geschehen: 

A="literal < 250 Byte" 

B="Literal < 250 Byte" 

C=A!!B 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0031I Name starts with numeric or "." 

Erklärung: Entsprechend der Regeln für Variablen darf der Name einer Variablen weder mit 
Punkt noch mit einer Zahl beginnen (man könnte sonst numerische Konstante redefinieren, 
was katastrophale Folgen hätte). 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Geben Sie der Variablen einen vernünftigen Namen. Auch wenn einige 
Sonderzeichen an Stelle 1 erlaubt sind, benutzen Sie einen Buchstaben. 

IRX0032I Invalid use of stem 

Erklärung: Ein Variablenstamm wurde auf falsche Weise genutzt (Das Konvertieren eines 
Variablenstammes über UPPER VAR. ist beispielsweise nicht möglich). 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Die Verarbeitung des Variablenstammes mit allen seinen Variablen muss 
notfalls in einer Programmschleife erledigt werden. 

IRX0033I Invalid expression result 

Erklärung: Das Resultat eines Ausdruckes ist falsch. Wurde beispielsweise in der Instruktion 
NUMERIC der Wert für DIGITS oder FUZZ verändert, kann es zu dieser Fehlermeldung kom-
men, weil der Wert von FUZZ nicht größer dem von DIGITS sein darf. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0034I Logical value not 0 or 1 

Erklärung: Ausdrücke in IF, WHEN, UNTIL oder WHILE müssen immer ein logisches 
TRUE(1) oder FLASE(0) liefern. Der Befehl: 

IF result THEN EXIT ist falsch, wenn der Inhalt von RESULT weder 0 noch 1 ist. Besser wäre 
in diesem Fall sicher  

IF result >< 0 THEN EXIT 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0035I Invalid expression 

Erklärung: Ein Formfehler in einem Ausdruck. Mögliche Ursachen sind zwei aufeinanderfol-
gende Bool'sche Operanden, ein Ausdruck, bei welchem hinter dem Vergleichsoperanden 
kein Wert folgt, oder ein Zuweisungsbefehl, bei dem in einer Konstanten nicht erlaubte Son-
derzeichen kodiert sind und keine Literalbegrenzer genutzt wurden (z.B.: var = 7 ++ b). 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0036I Unmatched "(" in expression 

Erklärung: Die Anzahl öffnender Klammern in einer Instruktion weicht von der Anzahl schlie-
ßender Klammern ab. Diese Situation tritt auf, wenn die Klammer in einem Befehl als Literal 
genutzt werden soll, aber keine Literalbegrenzer kodiert werden: 

Falsch:  IF POS((,variable) > 0 THEN ... 

Richtig : IF POS("(",variable) > 0 THEN... 

Hinweis:Prüfungen dieser Art finden nur für REXX-Instruktionen statt. Kodieren wir beispiels-
weise einen EXECIO-Befehl, bei dem die schließende Klammer nicht zwingend ist, ist das OK: 

Richtig : EXECIO 4 DISKR ddname (FINIS 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0037I Unexpected "," or ")" 

Erklärung: Komma wurde außerhalb eines Funktionsaufrufes (nicht als Fortsetzungszeichen) 
gefunden, oder in einem numerischen Ausdruck gibt es mehr schließende als öffnende Klam-
mern. Wird Komma in einem Character-Ausdruck benutzt, sollte dieser als Literal kodiert wer-
den: 

Falsch:  SAY Waehle A, B, oder C 

Richtig : SAY "Waehle A, B, oder C" 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0038I Invalid template or pattern 

Erklärung: Ein nicht erlaubtes Sonderzeichen wurde als Trennzeichen für Parsing genutzt 
(wie %). Die Meldung wird auch ausgegeben, wenn der Operand WITH in der Instruktion 
PARSE VALUE vergessen wird. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0039I Evaluation stack overflow 

Erklärung: Der Befehlsprozessor konnte einen Ausdruck aufgrund seiner Komplexität nicht 
auflösen (zu viele Klammern oder ineinander verschachtelte Funktionen). 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Brechen Sie den Ausdruck auf. Erzeugen Sie temporäre Variablen und füh-
ren Sie deren Ergebnisse wieder zusammen. 

IRX0040I Incorrect call to routine 

Erklärung: Häufigste Ursache hierfür ist, dass einer internen oder externen Funktion (REXX 
oder UDF) ein falscher Wert übergeben wurde. Dies ist abhängig von der jeweiligen Funktion. 

Sollte kein Funktionsaufruf stattfinden, liegt es oft an falscher Schreibweise. Wird ein Klam-
mernpaar im direkten Anschluss an ein Wort kodiert, geht der Befehlsprozessor immer davon 
aus, dass es sich hierbei um eine Funktion handelt. Deshalb: 

Falsch:  SAY Moegliche Eingabe(1/2) 

Richtig : SAY "Moegliche Eingabe(1/2)" 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0041I Bad arithmetic conversion 

Erklärung: Der Befehlsprozessor fand in einem numerischen Ausdruck einen Wert, der nicht 
numerisch war, oder einen Exponenten, der außerhalb der möglichen Werte (+999 999 999 
bis -999 999 999) liegt.Die häufigsten Ursachen sind Tippfehler im Namen einer Variablen 
oder TSO-Befehle, die nicht als Literale definiert wurden. Der Befehl EXECIO * DISKR führt 
zum Versuch, zwei Character-Variablen miteinander zu multiplizieren. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0042I Arithmetic overflow/underflow 

Erklärung: Eine numerische Variable hat den Maximal- oder Minimalwert überschritten (999 
999 999 bis -999 999 999). Häufige Ursache ist, wenn in einer Fehlerschleife eine Variable 
endlos mit sich selbst addiert oder multipliziert wird. Die Meldung wird auch ausgegeben, wenn 
mit 0 dividiert werden soll. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0043I Routine not found 

Erklärung: Im Programm erfolgte ein Funktionsaufruf oder eine CALL-Instruktion. Es gibt drei 
mögliche Ursachen: Der Name der Routine (Programm oder Funktion) wurde als Labelname 
im Programm nicht gefunden. Es handelt sich nicht um eine REXX-Funktion, und die externe 
Suche innerhalb der SYSEXEC/SYSPROC-Verkettung war ergebnislos. 

Sollte keine Funktion aufgerufen werden, ist die Meldung das Ergebnis einer falschen Schreib-
weise. Wird ein Klammernpaar unmittelbar an eine Zeichenfolge kodiert, geht der Befehlspro-
zessor immer davon aus, dass es sich hierbei um einen Funktionsaufruf handelt. Entweder 
haben Sie vergessen, ein Blank zwischen Zeichenfolge und Klammer zu kodieren, oder in 
einem arithmetischen Ausdruck fehlt ein Operationszeichen: 

Falsch:  ERG=VAR1(VAR2-7) 

Richtig : ERG=VAR1*(VAR2-7) 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0044I Function did not return data 

Erklärung: Eine Funktion liefert keine Daten zurück, obwohl sie dem Anschein nach fehlerfrei 
beendet wurde. Wird beispielsweise mit STORAGE(..) ein Speicherbereich gelesen, auf den 
man keinen Zugriff hat, gibt die Funktion keinen Wert zurück (STORAGE() ist in diesem Buch 
nicht dokumentiert). 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0045I No data specified on function RETURN 

Erklärung: Ein REXX-Programm wurde als Funktion aufgerufen und  bei dessen Programm-
ende wurden über die Anweisung RETURN keine Daten zurückgereicht. Eine Funktion muss 
immer einen Wert zurückreichen, da er im Ablauf des Programmes an die Stelle der Funktion 
tritt. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Korrigieren Sie Ihr Programm. 

IRX0046I Nicht belegt 

IRX0047I Nicht belegt 

IRX0048I Failure in system service 

Erklärung: Der Befehlsprozessor beendete ein Programm, weil eine Systemroutine (Ein-
/Ausgaben oder Stackmanipulation) mißglückte. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Überprüfen Sie Ihr Programm. Wenden Sie sich an die Systemprogram-
mierung. 

IRX0049I Interpreter failure 

Erklärung: Der Interpreter führt eine laufende Selbstprüfung durch. Endet eine dieser Prüfun-
gen mit einem schweren Fehler, wird diese Meldung ausgegeben. 

Systemreaktion: Programm-Abbruch. 

Bediener-Aktion: Informieren Sie die Systemprogrammierung. Leiten Sie eine Nachricht an 
IBM weiter. 



Inside z/OS Seite 168 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

2. Edit Makros 

Edit-Makros können genutzt werden, um den Editor über einen erweiterten Befehlssatz zu unterstüt-
zen. Grundsätzlich kann zwischen zwei Formen von Makros unterschieden werden: 

▪ einfache Makros, die eine Aneinanderreihung diverser Edit-Kommandos enthalten 
▪ erweiterte Makros, die sowohl Programmbefehle einer Sprache, als auch Befehle des Editors  

oder Dialog Manager Service enthalten. 

Um umfangreichere Makros zu erstellen, können Sprachen wie Fortran, Cobol oder PL/I, oder aber 
CLIST oder REXX genutzt werden. In der Praxis werden im allgemeinen CLIST oder REXX zur 
Kodierung erweiterter Makros genutzt.  

Im Weiteren werden die Erklärungen auf der Basis von REXX aufgesetzt. 

Edit Makros können auch Edit-Assignment-Statements enthalten, die zur Kommunikation zwischen 
dem Editor und dem Marco eingesetzt werden können. 

Die Makrosprache kennt für jeden Primär- oder Zeilenbefehl des Editors eine adäquate Instruktion. 
Daher kann das Makro generell alles, was ein Bediener im Editor kann, darüber hinaus aber auch 
noch einiges mehr. 

Edit Makros haben Zugriff auf die Services des Dialog Managers und des Systems. Da sie letztend-
lich Programme sind (oder Command-Lists) sind ihre Möglichkeiten nahezu grenzenlos. Die Gren-
zen eines Makros liegen in erster Linie dort, wo der der Programmierer an die seinen stößt. 

2.1. Anwendungsbereiche 

In der Praxis werden Edit Makros zu folgenden Zwecken eingesetzt: 

▪ wiederkehrende Befehlsgruppen zusammenfassen 
▪ Komplexe Anwendungen vereinfachen 
▪ Parameterübergaben 
▪ Informationsaustausch 

Die nachfolgenden Seiten zeigen für die einzelnen Anwendungen je ein kleines Beispiel. 

2.1.1. Wiederkehrende Befehle 

Das folgende Makro soll alle Zeilen in einer Adressdatei, bei denen die Postleitzahl nicht mit einer 8 
beginnt, löschen: 

/* MACDEMO1 ********************************/ 

ADDRESS ISREDIT "MACRO" 

ADDRESS ISREDIT "RESET EXCLUDED" 

ADDRESS ISREDIT "EXCLUDE ALL p'8####'" 

ADDRESS ISREDIT "FLIP" 

ADDRESS ISREDIT "DELETE ALL EXCLUDED" 

EXIT 0 

Die folgenden Screenshots zeigen eine Datei vor und nach Ausführung des Makros: 

Vor Befehlsausführung 

EDIT - WITTEMA.PDF.DATA(DEMO)  - LINE000 COLS 001 072 

COMMAND ==>  macdemo1                 SCROLL ==> PAGE 

****** ************** TOP OF DATA ******************* 

000001 HUGO   MUELLER   61333 NEBELHORN ADELESTR. 2 

000002 AUGUST BACHMAIER 85737 ISMANING  WASSERTURM 7 

000003 BRILLO HELMSTEIN 47111 FEUERBACH HALDE 47 

000004 MIKE   NAGEL     85737 ISMANING  ORFSTR. 9 

000005 HEINZ  DOOF      82319 STARNBERG JAKOBSTR.2B 

000006 RESI   HOLZAPFEL 12993 BANNWALD  ORTSRAND 58 

000007 ANTON  NIETNAGEL 55374 SCHILDA   BRUNFTGASSE 4 

****** ************* BOTTOM OF DATA ***************** 

Nach Befehlsausführung 

EDIT - WITTEMA.PDF.DATA(DEMO)  - LINE000 COLS 001 072 

COMMAND ==>  macdemo1                 SCROLL ==> PAGE 

****** ************** TOP OF DATA ******************* 

000001 AUGUST BACHMAIER 85737 ISMANING  WASSERTURM 7 

000002 MIKE   NAGEL     85737 ISMANING  DORFSTR. 9 

000003 HEINZ  DOOF      82319 STARNBERG JAKOBSTR.2B 

****** ************* BOTTOM OF DATA ***************** 

2.1.2. Vereinfachen komplexer Anwendungen 

Um komplexere Abläufe zu vereinfachen, kann ein Makro auch Logik enthalten. 

/* MACDEMO2 *********************************/ 

ADDRESS ISREDIT  

"MACRO" 

DO CNT=1 TO 3 

   "FIND 'ZEILE-?'" 

   IF RC = 0 THEN "CHANGE '?' '"cnt"'" 

   ELSE CNT=3 

END 

EXIT CODE(0) 

Das Beispiel bewirkt nachfolgende Veränderung in einer Datei: 

Vor Befehlsausführung 

EDIT - WITTEMA.PDF.DATA(DEMO)  - LINE000 COLS 001 072 

COMMAND ==>  macdemo1                 SCROLL ==> PAGE 

****** ************** TOP OF DATA ******************* 

000001 ZEILE-? 

000002 ZEILE-? 

000003 ZEILE-? 

000004 ZEILE-? 

000005 ZEILE-? 

****** ************* BOTTOM OF DATA ***************** 

Nach Befehlsausführung 

EDIT - WITTEMA.PDF.DATA(DEMO) -  LINE000 COLS 001 072 

COMMAND ==>                           SCROLL ==> PAGE 

****** ************** TOP OF DATA ******************* 

000001 ZEILE-1 

000002 ZEILE-2 

000003 ZEILE-3 

000004 ZEILE-? 

000005 ZEILE-? 

****** ************* BOTTOM OF DATA ***************** 

2.2. Parameterübergabe und Informationsaustausch 

Das folgende Beispiel zählt das Vorkommen des Strings ZEILE und teilt dies mit. 

/* REXX * MACDEMO3 *********************************/ 

ADDRESS ISREDIT "MACRO (PARM)" 

ADDRESS ISREDIT "SEEK ALL" PARM 

IF RC = 4 THEN DO 

   ZEDSMSG="Nicht gefunden" 

   ZEDLMSG="Zeichenfolge" PARM "wurde nicht gefunden" 

END 

ELSE DO 

   ADDRESS ISREDIT "(CNT) = SEEK_COUNTS" 

   ZEDLMSG="String" PARM ,"wurde" CNT "mal gefunden" 

END 

ADDRESS ISPEXEC "SETMSG MSG(ISRZ000)" 

EXIT 0 

Nach Aufruf des Makros wird die lange Nachricht ausgegeben. 

Vor Befehlsausführung 

EDIT - WITTEMA.PDF.DATA(DEMO) -  LINE000 COLS 001 072 

COMMAND ==>  macdemo1 zeile               SCROLL ==> PAGE 

****** ************** TOP OF DATA ******************* 

000001 ZEILE-1 

000002 ZEILE-2 

000003 ZEILE-3 

000004 ZEILE-? 

000005 ZEILE-? 

****** ************* BOTTOM OF DATA ***************** 

Nach Befehlsausführung 

EDIT - WITTEMA.PDF.DATA(DEMO) -  LINE000 COLS 001 072 

COMMAND ==>                           SCROLL ==> PAGE 

String ZEILE wurde insgesamt 5-mal gefunden 

000001 ZEILE-1 

000002 ZEILE-2 

000003 ZEILE-3 

000004 ZEILE-? 

000005 ZEILE-? 

****** ************* BOTTOM OF DATA ***************** 

2.3. REXX Makros 

REXX Edit Makros bestehen aus Befehlen, die in folgende Bereiche aufgeteilt werden können: 

▪ Edit Makro Befehle 
▪ REXX-Programmbefehle oder Kommentare 
▪ Dialog Manager Befehle 
▪ TSO-Kommandos 

Alle Statements werden vom Befehlsprozessor gelesen. Dabei werden symbolische Variable durch 
ihre Werte ersetzt. 

2.3.1. Edit Makro Kommandos 

müssen in REXX an das Environment ISREDIT übergeben werden. Nach Ersetzen der symboli-
schen Variablen durch ihre Werte werden die Kommandos vom Editor übernommen und ausgeführt. 

Beispiele für Edit Makro Befehle sind: 

ADDRESS ISREDIT "EXCLUDE ALL" 

ADDRESS ISREDIT "FIND 'zeichenfolge'" 

ADDRESS ISREDIT "RFIND" 

2.3.2. REXX - Programmbefehle 

übernehmen die Steuerung des Ablaufes des Makros. Befehle, die innerhalb eines Edit Makros häu-
fig vorkommen sind beispielsweise: 

variable=wert 

IF-THEN-ELSE-Anweisungen 

DO-????-END-Anweisungen 

EXIT-Anweisung 

2.3.3. Dialog Manager Services 

können innerhalb des Makros in Anspruch genommen werden, sofern die Adressierung an das En-
vironment ISPEXEC erfolgt. Häufig benutzte Dialog Manager Services sind: 

ADDRESS ISPEXEC "SETMSG... 

ADDRESS ISPEXEC "VPUT... 

ADDRESS ISPEXEC "DISPLAY... 

ADDRESS ISPEXEC "EDIT... 

ADDRESS ISPEXEC "LMCOPY... 

2.3.4. TSO-Kommandos 

Jeder Befehl, der vom REXX-Interpreter nicht ausgeführt werden kann und nicht an die Umgebun-
gen ISREDIT oder ISPEXEC adressiert wurde, wird zur Ausführung an das TSO weitergegeben. 
Um Probleme zu vermeiden, ist es sinnvoll, TSO-Kommandos als Literale zu hinterlegen. 

2.3.5. Programm Makros 

So wie Dialog Manager Dialoge in verschiedenen Programmiersprachen geschrieben werden kön-
nen, ist dies auch bei Edit Makros der Fall. Sprachen, die unterstützt werden sind: 

▪ PL/I 
▪ Cobol 
▪ Fortran 

▪ Pascal 
▪ APL/2 und 
▪ Assembler 

Da im weiteren Verlauf nur auf REXX als unterstützende Sprache eingegangen wird, verzichten wir 
hier auch auf die Beschreibung der Unterschiede zwischen REXX/CLIST- und Programm-Makros. 
Details können nachgelesen werden im IBM-Manual "Edit und Edit Makros". 

2.4. Regeln für Edit Makros 

Befehle, die in der Kommandozeile des Editors (Primary Commands) eingegeben werden können, 
können auch in Makros genutzt werden. Bei einigen Befehlen gibt es aber Unterschiede, ob sie 
direkt unter Edit, oder innerhalb eines Makros genutzt werden: 

▪ Wird ein Befehl direkt unter Edit genutzt, sind dessen Ergebnisse (O.K.- oder Fehlermeldung) 
immer unmittelbar sichtbar. Wird der Befehl im Makro kodiert, werden keine Meldungen ge-
zeigt und die aktuelle Darstellung des Bildschirmes (sichtbare Zeilen) kann unterschiedlich 
sein. 

▪ Werden in einem Makro mehrere Befehle hintereinander ausgeführt, ist dies im Einzelnen 
am Terminal nicht sichtbar. Als Ergebnis wird der Zustand der Daten nach dem letzten aus-
geführten Befehl gezeigt. 

▪ Manche Befehle können im Makro mit zusätzlichen Operanden versehen werden, die direkt 
im Editor nicht unterstützt werden. 

Darüber hinaus sollten bereits bei Erstellung eines Makros einige Dinge beachtet werden: 

2.4.1. Namensvergabe 

Es kann jeder beliebige alphanumerische Name bis acht Byte Länge genutzt werden. Stelle Eins 
des Namens darf nicht numerisch sein. Nutzt sprechende Namen. 

2.4.2. Variablen 

Variablen, die in einem Edit-Makro genutzt werden, müssen die gleichen Anforderungen erfüllen wie 
in ISPF.  

Wenn keine Variableninterpretation gewünscht ist (weil der Name der Variablen übergeben wird, 
nicht ihr Inhalt), werden Variablen zwischen runde Klammern gesetzt und auf das führende Amper-
sand(&) verzichtet. 

2.4.2.1. Variablenverarbeitung 

Der SCAN-Befehl steuert den Austausch von Variablen durch ihre Werte in den Makrobefehlen, 
bevor sie an den Editor übergeben werden. Ist der Scanmodus aktiv, werden Makrobefehle nach 
Ampersands durchsucht. Wird ein Ampersand unmittelbar von einem Zeichen ungleich Blank ge-
folgt, wird dies als der Name der Variablen interpretiert. Der Name muss durch Blank oder Punkt 
abgeschlossen werden. Der Punkt erlaubt die Verkettung von Variablen ohne Zwischenräume. Dies 
sollte berücksichtigt werden, wenn Programm-Makros kodiert werden, die keinen Interpreter im Hin-
tergrund haben. 

2.4.2.2. Variablenzuweisung 



Inside z/OS Seite 173 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

Über die Variablenzuweisung können Editor und Makro kommunizieren. Der Befehl besteht immer 
aus zwei Teilen, einem Schlüsselwort und einem Wert. Das Schlüsselwort repräsentiert Daten in-
nerhalb des Editors, der Wert die Daten im Makro. Auf diese Weise können Daten zwischen Makro 
und Editor in beiden Richtungen ausgetauscht werden. 

Die Zuweisung erfolgt von rechts nach links. Der Ausdruck (rechts von =) wird aufgelöst, das Ergeb-
nis wird der Variablen (links von =) zugewiesen. 

ADDRESS ISREDIT "keyword = (variable)" 

ADDRESS ISREDIT "(variable)=keyword" 

2.4.2.3. Mögliche Werte 

Der Wert einer Variablenzuweisung im Makro (steht rechts vom Gleichheitszeichen) kann folgen-
dermaßen aussehen: 

▪ ein Literal. Einfache Literale sind eine beliebige Folge von Zeichen ungleich Blank. Abge-
grenzte Literale werden von Apostrophen(') oder Quotes(") eingeschlossen. 

▪ eine Dialogvariable, eingeschlossen in runde Klammern 

Wird die Variable rechts vom Gleichheitszeichen genutzt, wird für die Wertzuweisung ihr gesamter 
Inhalt, inklusive sämtlicher Quotes, Apostrophe und Blanks genutzt. 

Im Umfeld von REXX sind Function- und Shared-Pool vermischt. In einem Progamm-Makro muss in 
jedem Fall mit VGET/VPUT-Anweisungen gearbeitet werden. 

2.4.2.4. Schlüsselwort-Angaben 

können aus einem einfachen Schlüsselwort bestehen, oder aus einem Schlüsselwort, gefolgt von 
Zeilennummer oder Marke (Label). Die folgenden Beispiele verdeutlichen die unterschiedlichen 
Möglichkeiten: 

ADDRESS ISREDIT "CURSOR=2,25" 

ADDRESS ISREDIT "CURSOR=2 25" 

Beide Formen der Zuweisung sind möglich. Wird mehr als ein Wert zugewiesen, sind die einzelnen 
Werte durch Komma oder Blank zu trennen. Werden numerische Werte übergeben, dürfen Apostro-
phe oder Quotes nicht genutzt werden (Quotes, die den Makrobefehl eingrenzen werden vom Inter-
preter bereits aufgelöst). Das nachfolgende Beispiel ist falsch: 

ADDRESS ISREDIT "CURSOR='2','25'" 

ADDRESS ISREDIT "CURSOR='2,25'" 

Sollen mehrere Variable gefüllt oder ausgelesen werden, sieht die entsprechende Syntax folgender-
maßen aus: 

ADDRESS ISREDIT "(var1,var2)=wert1 wert2" 

ADDRESS ISREDIT "var1 var2=(var3 var4)" 

Zur Trennung von Variablen kann Komma oder Blank genutzt werden. Sollen führende Parameter 
übergangen werden, sind entsprechend führende Kommata zu kodieren: 

ADDRESS ISREDIT "(,var2)=wert1 wert2" 

ADDRESS ISREDIT "var1 var2=(,var4)" 

2.4.2.5. Anwendung der Variablenzuweisung 

Das folgende Beispiel zeigt eine gängige Situation für eine Variablenzuweisung im Makro. Während 
einer ISPF-Sitzung kennt der Editor die Einstellungen aller Modi aus dem Profil. Das Makro benötigt 
diese Informationen aber ebenfalls. Beschränken wir uns für das folgende Beispiel auf den Zustand 
des Schalters CAPS. 

"(AKTCAPS) = CAPS" 

Zwischen runden Klammern steht der Name der Variablen, in der der CAPS-Modus (ON oder OFF) 
hinterlegt wird. Erkennt der Editor einen zwischen Klammern eingeschlossenen Variablennamen an 
einer Stelle, an der eigentlich ein Wert erwartet wird, erfolgt vor Verarbeitung eine Variablensubsti-
tution. Wenn wir davon ausgehen, dass bei obiger Situation CAPS auf ON gesetzt war und dieser 
Status wiederhergestellt werden soll, kann eine der nachfolgenden Schreibweisen genutzt werden: 

"CAPS = ON" 

"CAPS = (AKTCAPS)" 

"CAPS = &AKTCAPS" 

"CAPS = "AKTCAPS 

Im letzten Beispiel ersetzt der REXX-Interpreter die Variable durch ihren Inhalt, bevor der Befehl an 
den Editor zur Ausführung übergeben wird. 

Im vorletzten übergibt der REXX-Interpreter das Literal unverändert an den Befehlsprozessor von 
ISPF. Dieser ersetzt die Variable &AKTCAPS. 

Im zweiten Beispiel läuft ein impliziter VGET-Befehl ab, über den der Inhalt der Variablen AKTCAPS 
aus dem Shared-Pool gelesen wird. 

Einige Informationen können zwischen Makro und Editor paarweise in beiden Richtungen ausge-
tauscht werden.  

"(LBND,RBND) = BOUNDS" 

Die linke und rechte Boundarygrenze wird in den Variablen LBND und RBND abgelegt. 

"BOUNDS = &LBND,&RBND" 

Die linke und rechte Boundarygrenze wird entsprechend der Werte der Variablen LBND und RBND 
gesetzt. 

Im Kapitel "Edit Makro Befehle" kann nachgelesen werden, welche Befehle mit einer oder zwei Va-
riablen genutzt werden können. 

2.4.2.6. Datenmanipulation durch Variablenzuweisung 

Variablenzuweisung kann auch eingesetzt werden, um Zeileninhalte zu sichern, zu verändern oder 
zu ergänzen. Sehen wir uns dazu die folgenden Befehle an: 

ADDRESS ISREDIT "TEMPLINE = LINE 3" 

Kopiert den Inhalt der dritten Zeile der Datei in die Variable TEMPLINE. 

ADDRESS ISREDIT "LINE_AFTER 7 = LINE 3" 

Kopiert den Inhalt der Zeile 3 hinter die Zeile 7 

ADDRESS ISREDIT "LINE 9 = (TEMPLINE)" 

Verändert den Inhalt der Zeile 9 und überschreibt ihn mit dem Inhalt der Variablen TEMPLINE. 

ADDRESS ISREDIT "LINE_AFTER 4 = (NEWLINE)" 

Sorgt für das Einfügen einer neuen Zeile hinter der Zeile 4. Bei NEWLINE handelt es sich um eine 
Variable, deren Inhalt für den INSErT benutzt wird. 

2.4.2.7. Unterschiede zwischen Edit- und REXX-Zuweisungen. 

Folgende Unterschiede müssen berücksichtigt werden: 

▪ Edit-Zuweisungen müssen immer an ISREDIT adressiert werden. CLIST Zuweisungen erfol-
gen durch den SET-Befehl, REXX-Zuweisungen in der Schreibweise var=wert. 

▪ In einer Edit-Zuweisung muss auf der linken oder rechten Seite des Gleichheitszeichens ein 
Schlüsselwort (eventuell gefolgt von einer Zeilennummer oder einem Label) stehen. 

▪ Werden in Edit-Zuweisungen Variable eingesetzt, werden die Namen der Variablen zwischen 
Klammern und ohne führendes Ampersand geschrieben. In einigen Fällen können auch zwei 
Variablennamen in der Klammer kodiert werden. 

▪ Rechenoperationen sind in einer Edit Zuweisung nicht erlaubt. In speziellen Fällen kann 
durch das Pluszeichen(+) für eine teilweise Überlagerung des Zeileninhaltes gesorgt werden. 

2.4.2.8. Überlagerungen 

Bei der Wertzuweisung kann eine komplexe Kombination mehrerer Variablen und Literale genutzt 
werden. Diese Form nennt man eine Überlagerung (Overlay). Wenn Overlays ausgeführt werden, 
sollten einige Dinge beachtet werden: 

Werden zwei Angaben auf einer Seite des Gleichheitszeichens durch Plus(+) verbunden, führen die 
Zeichen rechts vom Plus zur Überlagerung der Angaben links vom Gleichheitszeichen, sofern diese 
Blanks sind. Sehen wir uns hierzu zwei Beispiele an: 

"LINE .ZCSR = LINE + '//'" 

In der Zeile, in der der Cursor steht, werden die ersten beiden Stellen durch // ersetzt. Der Rest der 
Zeile bleibt unverändert. 

"MASKLINE=MASKLINE + <40 '/*' 70 '*/'>" 

Die Spalten 40/41 und 70/71 der aktuellen Maskenzeile werden jeweils mit Kommentarbeginn und -
ende ergänzt. 

In beiden Fällen wird die Überlagerung nur durchgeführt, sofern die Stellen im Zielbereich Blank 
oder deckungsgleich sind. 

Im letzten Beispiel wurde eine Schablone angelegt, die folgendes Format aufweist: 

<spalte1 literal1 spalte2 literal2 .... > 

Die Kodierung erfolgt im Wechsel zwischen Spaltenangabe und Literal, das an der angegebenen 
Spalte positioniert wird. Die gesamte Schablone wird von den Zeichen Kleiner(<) und Größer(>) 
eingeschlossen. Spalte und/oder Literal können auch über Variable geliefert werden. Die Variablen 
müssen dabei in runde Klammern eingeschlossen werden. Daraus resultierende Kodierungen sehen 
folgendermaßen aus: 

<(s-var1) (d-var1) (s-var2) (d-var2) ... > 

<(s-var1,d-var1) (s-var2,d-var2) ....... > 

<(s-var1) literal1 spalte2 (d-var2) .... > 

Beim Einsatz von Variablen dürfen die Namen maximal achtstellig sein. 

2.4.3. Ausführung von Zeilenkommandos 

Zeilenbefehle wie M(ove) können unmittelbar vom Makro aus nicht genutzt werden. Viele Zeilen-
kommandos können in Makros aber über Primärbefehle oder Variablenzuweisungen nachvollzogen 
werden. Auf diese Weise sind Aktionen wie Zeilenkommandos C(opy), M(ove) oder R(epeat) mög-
lich. Um beispielsweise das Zeilenkommando M(ove) in einem Makro nachzuvollziehen, wird der 
Zeileninhalt in einer Variablen hinterlegt. Die Zeile kann anschließend gelöscht und der Inhalt der 
Variablen hinter einer anderen Zeile eingefügt werden. 

2.4.4. Parameterübergabe an ein Makro 

Parameter können simple oder zwischen Apostrophe eingegrenzte Zeichenketten sein. Die Über-
gabe kann entweder über die VPUT/VGET-Services des Dialog Managers erfolgen, oder, wie bei 
den Edit-Primärkommandos, durch die Angabe hinter dem Makronamen. 

Sind in einem Makro Parameter vorgesehen und werden diese bei Aufruf des Makros nicht überge-
ben, führt dies zum Fehler. Wurde das Makro als Kommando von der Tastatur aus aufgerufen, gibt 
der Editor eine entsprechende Meldung aus. Die Prüfung, ob Parameter aber in der richtigen Anzahl 
und Reihenfolge angegeben wurden, ist Sache des Programmierers. 

Sollen Parameter an ein Makro übergeben werden, müssen diese in der Kopfzeile des Makros, zwi-
schen runden Klammern eingeschlossen, angegeben werden. Durch den Aufruf: 

MACDEMO4 HUBER 

und eine entsprechende Kodierung: 

ADDRESS ISREDIT "MACRO (SUCH)" 

wird der Parameter HUBER im Makro in der Variablen SUCH hinterlegt. 

Werden mehrere Parameter beim Aufruf angegeben, erfolgt die Zuweisung folgendermaßen: 

Sind in der Kopfzeile des Makros drei Variable definiert und werden beim Aufruf drei Parameter 
angegeben, erfolgt die Zuweisung der Reihe nach von links nach rechts. Der erste Parameter wird 
der ersten Variablen im Makro zugewiesen, und so weiter. 

Sind mehr Variable im Makro definiert, als Parameter übergeben werden, werden die überzähligen 
Variablen auf Nullstring gesetzt. 

Werden mehr Parameter angegeben, als Variable vorgesehen sind, erfolgt die Zuordnung der Reihe 
nach, wobei die überzähligen Parameter als Zeichenkette in der letzten Variablen vereint werden.  

Wird dementsprechend nur eine Variable im Makro definiert und beim Startaufruf eine Reihe von 
Parametern angegeben, liegen diese gesammelt in der entsprechenden Variablen vor. 

Parameter werden durch Blank oder Komma getrennt. Ein Parameter kann aber auch eine beliebige 
Zeichenfolge (mit Blanks und/oder Kommas) sein, sofern sie zwischen Quotes oder Apostrophe ein-
geschlossen ist. Werden mehrere Parameter in Form abgegrenzter Zeichenketten übergeben, sind 
diese wiederum durch Blank oder Komma voneinander zu trennen. Betrachten wir das Ganze in 
einem Beispiel. Angenommen, im Makro ist folgende Kopfzeile kodiert 

ADDRESS ISREDIT "MACRO (VAR1,VAR2,REST)" 

und der Aufruf erfolgt durch 

MACDEMO5 DER PFERD HAT VIER BEIN 

ist das Ergebnis: 

VAR1 DER 

VAR2 PFERD 

REST HAT VIER BEIN 

Erfolgt der Aufruf dagegen in der Form 

MACDEMO5 'DER PFERD' 'HAT VIER BEIN' 

ist das Ergebnis: 

VAR1 DER PFERD 

VAR2 HAT VIER BEIN 

REST nullstring 

2.4.5. Meldungsausgabe mit Makros 

Meldungen werden auf die gleiche Art erzeugt, wie in einem ISPF-Dialog. 

Über  SETMSG kann auf eine Meldung gezeigt werden, die mit der nächsten Panelaktion sichtbar 
wird. Mit dem DISPLAY-Service in Verbindung mit dem Schlüsselwort MSG(..). Dieses Verfahren 
empfiehlt sich, wenn das Makro seinerseits mit Panels arbeitet. 

Unter ISPF sind drei Meldungen vorbereitet, die ohne eigene MSG-Library genutzt werden können. 
Die kurze und lange Nachricht kann zu gegebener Zeit in Form von Variablen vorbereitet werden. 

 ISRZ000 '&ZEDSMSG' .ALARM=NO .HELP=ISR2MACR 

'&ZEDLMSG' 

ISRZ001 '&ZEDSMSG' .ALARM=NO .HELP=ISR2MACR 

'&ZEDLMSG' 

ISRZ002 '&ZEDSMSG' .ALARM=NO .HELP=ISR2MACR 

'&ZEDLMSG' 

Soll in einem Makro die Meldung "PARAMETER FALSCH" oder "EINGABE MUSS NUMERISCH 
SEIN" ausgegeben werden, bedingt dies im Makro: 

ZEDSMSG="PARAMETER FALSCH" 

ZEDLMSG="EINGABE MUSS NUMERISCH SEIN" 

ADDRESS ISPEXEC "SETMSG MSG(ISRZ000)" 

2.5. Labels in einem Makro 

Labels werden genutzt, um eine Datenzeile zu benennen. Diese Namen bleiben immer fest mit der 
Zeile verbunden. Sie beginnen mit Punkt(.), gefolgt von maximal 7 Alphazeichen. Das erste darf kein 
Z sein. Z-Label sind dem Editor selbst vorbehalten. 

Labelnamen und ihre Bedeutung, wie sie im Editor Standard sind. 

Labelname Bedeutung 

.ZCSR Datenzeile, in der der Cursor steht 

.ZFIRST Erste Datenzeile (kann .ZF abgekürzt werden) 

.ZLAST Letzte Datenzeile (mögliches Kürzel .ZL). 



Inside z/OS Seite 178 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

.ZFRANGE Die erste Zeile eines vom Benutzer definierten Berei-
ches 

.ZLRANGE Die letzte Zeile eines vom Benutzer definierten Berei-
ches 

.ZDEST Zielzeile, definiert durch den Anwender 

 

Die Label .ZCSR, .ZFIRST und .ZLAST sind nicht zeilengebunden. .ZCSR ist cursorabhängig. 
.ZFIRST zeigt immer auf die erste, .ZLAST immer auf die letzte Zeile. 

2.5.1. Anwendung der Label 

Im Makro erfolgt die Festlegung eines Labels durch den Zuweisungsbefehl. Label können dort auch 
statt der Zeilennummer eingesetzt werden. 

Gehen wir davon aus, dass innerhalb eines bestimmten Bereiches eine Verarbeitung stattfinden soll. 
Die aktuelle Zeile, in der der Cursor steht, soll mit dem Label BEGINN versehen werden. 

ADDRESS ISREDIT "LABEL .ZCSR = .BEGINN" 

Soll vor dieser Zeile eine Kommentarzeile eingefügt werden, kann das wie folgt aussehen: 

CM="/** START" COPIES("*",60)"/" 

ADDRESS ISREDIT "LINE_BEFORE .BEGINN = &CM" 

Oder 

ADDRESS ISREDIT "LINE_BEFORE .BEGINN = (CM)" 

Oder 

ADDRESS ISREDIT "LINE_BEFORE .BEGINN =" CM 

Ist ein Label bereits für eine andere Zeile vergeben, "zieht es um", sofern die Vergabe des Labels 
auf der gleichen hierarchischen Ebene stattfindet. 

Wird mit dem Label .ZCSR gearbeitet, ist zu bedenken, dass  die Befehle FIND, CHANGE, 
EXCLUDE, SEEK, TSPLIT und CURSOR die Position des Cursors verändern. 

Wird einer Zeile ein neues Label gegeben, werden alte Label überschrieben, sofern sich das Ganze 
in der gleichen Hierarchieebene abspielt. Dies gilt auch für das Zurücksetzen eines Labels in der 
Form: 

ADDRESS ISREDIT "LABEL .BEGINN = ' '" 

Über die LINENUM-Zuweisung kann abhängig vom Return Code festgestellt werden, ob ein Label 
vorhanden ist, oder nicht. 

ADDRESS ISREDIT "(TMP) = LINENUM .BEGINN" 

IF RC = 8 , 

 THEN /* Label nicht vorhanden */ 

 ELSE /* Label vorhanden */ 

Details über die Abfrage der Return Codes können im "Band-2 Syntax Referenz" nachgelesen wer-
den. 

2.5.2. Bezugnahme auf Label 

Makros können, ob sie ineinander verschachtelt sind oder nicht, auf Label höherer Hierarchiestufen 
zugreifen. Label werden der Hierarchiestufe zugewiesen, aus der sie gebildet wurden und können 
problemlos gleichnamig zu Labels anderer Hierarchiestufen sein. Endet ein Makro, werden alle von 
ihm gebildeten Label aufgelöst. Label und Hierarchiestufe können über einen Zuweisungsbefehl er-
mittelt werden: 

ADDRESS ISREDIT "(LAB,HIER) = LABEL (zeile)" 

In der ersten Variablen wird der Labelname, in der zweiten die Hierarchiestufe, beides bezogen auf 
die angegebene Zeilennummer des Labels, hinterlegt. Ist kein Label für diese Zeile gesetzt, werden 
beide Variablen auf Nullstring gesetzt. 

Label können auch außerhalb der jeweiligen Hierarchiestufe erzeugt werden: 

ADDRESS ISREDIT "LABEL zeile  = label (level)" 

Wird ein Label in einer höheren Hierarchiestufe gesetzt, das dort bereits definiert war, wird es über-
schrieben und bleibt solange gültig, bis es aufs Neue überschrieben wird, oder die betreffende Hie-
rarchiestufe aufgelöst wird. 

Label, die ohne Angabe einer Hierarchiestufe, oder mit tieferer Hierarchiestufe als die aktuelle ge-
setzt werden, werden in der aktuellen Hierarchiestufe gebildet. Dies hat keine Auswirkung auf Labels 
höherer Hierarchiestufen. 

2.5.3. Hierarchiestufen 

Jedes Makro arbeitet auf einer eigenen Hierarchiestufe, für die jeweils ein eigenes Umfeld, ver-
gleichbar zu den Functionpools im Dialog Manager, geschaffen wird. Der Terminalbediener, der in-
teraktiv mit dem Editor arbeitet, befindet sich in der obersten  hierarchischen Ebene (Stufe 0). ruft 
der Anwender ein Makro auf, läuft dieses auf Stufe 1. Ruft das Makro wiederum ein Makro, wird 
innerhalb der Hierarchie jeweils eine Stufe nach unten verzweigt. Aus der jeweils aktuellen Hierar-
chiestufe kann die Verschachtelungstiefe der Makros abgelesen werden. 

Wenn ein Makro ohne Levelangabe nach einem Label fragt, oder ein Label als Zeilenzeiger benutzt, 
wird das Label zunächst innerhalb der aktuellen Hierarchiestufe gesucht. Ist es dort nicht vorhanden, 
wird die Suche innerhalb der Hierarchie nach oben stufenweise fortgesetzt, bis das Label gefunden 
wird. 

Versucht ein Makro ein Label außerhalb der Hierarchiestufe zu setzen, wird folgendermaßen ver-
fahren: 

• Wird ein Wert für die Hierarchiestufe kleiner als 0 angegeben, wird 0 angenommen. 

• Wird ein Wert größer der aktuellen Stufe angegeben, wird das Label innerhalb der aktuellen 
Stufe gesetzt. Es können keine Label für aufzurufende Makros vorbereitet werden. 

Über folgenden Befehl ist ein Makro in der Lage, die aktuelle Hierarchiestufe selbst zu ermitteln: 

ADDRESS ISREDIT "(variable) = MACRO_LEVEL" 

2.6. Bezugnahme auf Datenzeilen und Spalten 

Datenzeilen können über ihre relative Zeilennummer oder über Labels angesprungen werden. Spe-
zielle Zeilen wie MASK, COLS, BNDS, TABS oder MSG sind keine Datenzeilen. Ausgeblendete 
Zeilen hingegen werden als Datenzeilen betrachtet. 

Relative Zeilennummern sind unabhängig vom Nummerntyp. Die Hinweiszeile "TOP OF DATA" trägt 
immer die relative Nummer 0, die erste Datenzeile immer die Nummer 1, und so weiter. 

Beim Einfügen oder Löschen von Datenzeilen verändern sich ab der betreffenden Zeile nach unten 
alle Zeilennummern. 

Im Makro ist nur der editierbare Teil der Datenzeile verfügbar. Wenn "NUMBER ON" gesetzt ist, wird 
die Zeilennummer ausgegrenzt (bei "NUMBER ON COB" sind die ersten sechs Byte, bei "NUMBER 
ON" die letzten acht Byte nicht editierbar). Im Nummernmodus für Cobol wird der absoluten Spalte 
7 die relative Spaltennummer 1 zugewiesen. Werden Spaltenangaben von einem Bediener bei Auf-
ruf des Makros mitgeliefert, sind diese mit Vorsicht anzugehen. Im Allgemeinen müssen diese An-
gaben gemäß dem Nummernmodus konvertiert werden. 

Sollen die Zeilennummern als Daten behandelt werden, muss das Makro den aktuellen Nummern-
modus speichern, den Modus "NUMBER OFF" einschalten und nach Verarbeitung den alten Zu-
stand wieder herstellen. 

Will ein Makro die aktuelle Spaltenposition des Cursors ermitteln, wenn dieser außerhalb der edi-
tierbaren Datenzeile steht, wird eine 0 übergeben. Wird die Spaltenposition auf 0 gesetzt, wird der 
Cursor in das Zeilenkommandofeld der aktuellen Zeile gesetzt. 

2.7. Makro-Definitionen 

Über den DEFINE-Befehl können folgende Aktionen durchgeführt werden: 

▪ Aliasnamen für Edit-Primärbefehle vergeben 
▪ Edit-Befehle durch eigene ersetzen 
▪ Makros als Programm-Makros identifizieren 
▪ Befehle ungültig setzen 

Details können im Kapitel "Edit Makro Befehle" nachgelesen werden. 

2.7.1. Definition von Aliasnamen 

Um einen Aliasnamen für einen Befehl festzulegen, wird erst der Aliasname, dann das Schlüsselwort 
ALIAS und zuletzt der originäre Name des Befehles kodiert. 

ADDRESS ISREDIT "DEFINE S ALIAS SAVE" 

Erlaubt zur Sicherung der bearbeiteten Datei die Eingabe "S" anstelle von SAVE. Generell könnte 
hier auch die Edit-Command-Table bemüht werden. 

2.7.2. Definitionen rücksetzen 

Soll die oben angeführte Definition wieder zurückgenommen werden, ist folgende Kodierung nötig: 

ADDRESS ISREDIT "DEFINE S RESET" 

2.7.3. Ersetzen von Edit-Befehlen 

Um einen Befehl des Editors durch ein Makro zu ersetzen, wird einer der nachfolgenden Befehle 
kodiert: 

ADDRESS ISREDIT "DEFINE COPY MACRO PGM CPY" 

ADDRESS ISREDIT "DEFINE COPY MACRO CMD CPY" 

Der obere Befehl muss angewandt werden, wenn es sich um ein Programm-Makro handelt. Ist das 
Makro in CLIST oder REXX kodiert, ist die zweite Form anzuwenden. 

Um die Definition wieder rückgängig zu machen, wird kodiert: 

ADDRESS ISREDIT "DEFINE COPY RESET" 

Soll im Makro der Built-in-Befehl COPY benutzt werden, obwohl er per Define umngelenkt wurde, 
ist zu kodieren: 

ADDRESS ISREDIT "BUILTIN COPY" 

2.8. Suchfolge für Makroaufrufe 

Wird im Editor oder Makro ein unbekannter Befehl abgesetzt, wird die SYSPROC/SYSEXEC-Ver-
kettung nach einem gleichnamigen Member durchsucht. Wird eines gefunden, wird automatisch da-
von ausgegangen, dass es sich um ein Makro handelt. 

Wird ein Programm-Makro aufgerufen, wird stellvertretend die Verkettung von ISPLLIB, STEPLIB 
und LINKLST durchsucht. Bei Programm-Makros muss der Makroname mit einem Ausrufezeichen(!) 
beginnen. Für den Namen stehen somit nur maximal sieben Byte zur Verfügung. 

2.9. Anwendung von PROCESS 

Durch den PROCESS-Befehl kann die reguläre Abarbeitung der Makrobefehle verändert werden. 
Der Befehl steht in Verbindung zur Schlüsselwortangabe in der Kopfzeile des Makros.  

PROCESS bewirkt, dass Veränderungen im Full-Screen-Modus und Zeilenbefehle vor Interpretation 
von Primärkommandos ausgeführt werden. 

Bei NOPROCESS wird die Veränderung editierbarer Daten, wie die Ausführung von Zeilenkomman-
dos, zurückgehalten, bis später im Makro eine PROCESS-Anweisung erkannt, oder das Makro be-
endet wird.  

Die Angabe NOPROCESS kann aus zweierlei Gründen sinnvoll sein: 

Befehle sollen ausgeführt werden, bevor Veränderungen im editierbaren Teil oder Zeilenkomman-
dos ausgeführt werden. 

Im Makro wird ein PROCESS-Befehl kodiert, um festzulegen, wie Zeilenkommandos, die einen Be-
reich, eine einzelne Zeile, oder beides beschreiben, gehandhabt werden sollen. Auf diese Weise ist 
eine Prüfung eingegebener Zeilenkommandos sinnvoll möglich (Edit REPLACE arbeitet auf diese 
Weise). 

2.9.1. Bereichsdefinitionen durch den Bediener 

Um dem Anwender die Möglichkeit zur Eingabe von Block-Zeilenbefehlen zu geben, muss im Makro 
folgende Zeile kodiert sein: 

ADDRESS ISREDIT "PROCESS RANGE argument" 

Das Argument kann aus einem oder zwei  Befehlen bestehen, die  akzeptiert werden. Wird beispiels-
weise kodiert: 

ADDRESS ISREDIT "PROCESS RANGE Y" 

kann der Bediener einen Zeilenbefehl in folgender Weise einsetzen: 

▪ Y auf eine einzelne Zeile 
▪ YY als Blockbefehl (paarweise) 
▪ Yn wobei n die Anzahl Zeilen ist, die von dem Befehl betroffen werden. 

PROCESS erlaubt die Angabe von maximal zwei Zeilenbefehlen. Wird dies genutzt, kann über 
RANGE_CMD erkannt werden, welcher der beiden Befehle getippt wurde. Der Name der Zeilen-
kommandos kann bis zu sechs Byte lang sein (Blockkommandos sind dann aber nicht mehr mög-
lich). Die Zeichen BLANK, Minus(-) und Apostroph(') dürfen nicht benutzt werden. 

Nachdem der PROCESS-Befehl ausgeführt ist, werden die Variablen .ZFRANGE auf die erste rela-
tive Zeilennummer und .ZLRANGE auf die letzte relative Zeilennummer des ausgewählten Berei-
ches (kann die gleiche wie .ZFRANGE sein) gesetzt. 

Wurde kein Bereich durch Zeilenkommandos ausgewählt, werden die beiden Variablen auf die erste 
und letzte Datenzeile gesetzt und RC weist darauf hin, dass keine Bereiche angegeben wurden. 

2.9.2. Zielangaben durch den Bediener 

Wird im Makro  

ADDRESS ISREDIT "PROCESS DEST" 

kodiert, wird dem Bediener eine Zielangabe in Form der Zeilenkommandos A/B (After oder Before) 
erlaubt. Die Variable .ZDEST enthält in diesem Fall die relative Zeilennummer, in der die Zielortan-
gabe steht. 

Sind weder A noch B angegeben, wird .ZDEST auf die letzte Datenzeile gesetzt und über einen 
Return Code mitgeteilt, dass die Zielortangabe fehlte. 

2.10. Recovery Macros 

Tritt während einer Edit-Sitzung ein Systemfehler auf, wird dem Anwender bei Wiederanlauf ein 
Recovery-Menü angeboten. Hier kann, so gewollt, ein Recovery Makro zum Einsatz kommen, wel-
ches alle Befehlsdefinitionen, Aliasnamen etc. wiederherstellt. Um diesen Service zu aktivieren muss 
in einem Recovery Macro folgende Zeile kodiert sein: 

ADDRESS ISREDIT "RMACRO = macroname" 

macroname ist der Name des Recovery Makros. 

2.11. Fehlerbehandlung und Test 

Grundsätzlich gilt: Am besten ist es, keine Fehler zu machen. Wenn sich aber Arbeit nicht vermeiden 
lässt, sind die ersten Resultate häufig nicht so, wie man es sich wünscht. Auch einem Makro-Insider 
unterlaufen Fehler, die zum Abbruch des Makros führen, oder das Makro läuft anders, als gewollt. 
In diesem Fall ist es am vernünftigsten, ein Makro im Ablauf zu verfolgen, wie man dies von vielen 
Sprachen her kennt (debugging). 

Nachdem in einem Makro unterschiedlichste Elemente vereint werden können, sind auch die Varia-
tionen im Ärgernisbereich äußerst facettenreich. 

Grundsätzlich können in umfangreicheren Makros folgende Quellen die Ursache für ungewollte Ab-
läufe sein: 

• Edit-Befehle 

• Makro Befehle 

• Dialog Management Services 

• Sprachelemente der Funktion (CLIST, REXX etc.) 

• TSO-Befehle 

Und nicht zu vergessen: Ein Makro macht immer was man kodiert - nie was man will! 

Die nachfolgenden Seiten sollen eine kleine Hilfestellung bieten und zeigen, auf welche Arten Fehler 
in einem Makro, oder in dessen logischem Ablauf, erkannt und behoben werden können. 

2.11.1. Fehler in Edit-Befehlen 

Fehler in den Edit-Befehlen werden vom Editor selbst erkannt. Sie werden entweder über ein Edit-
Makro-Fehlerpanel gemeldet, das eine Fehlerbeschreibung enthält, oder über einen entsprechen-
den Return Code. Bezugnehmend auf Edit Fehler sollten einige Dinge bedacht werden: 

• Wenn der ISPF-Testmodus ausgeschaltet ist, endet das Makro abnormal mit einem Rück-
sprung in die Edit-Session. 

• Ist der ISPF-Testmodus aktiv, arbeitet auch PDF im Testmodus. Ist zudem "CONTROL ER-
RORS CANCEL" gesetzt (Defaultwert), wird bei einem Fehler ein entsprechendes Panel ge-
zeigt, über das durch die Angabe "YES" eine Fortsetzung des Ablaufes erreicht werden kann. 
Wurde im Makro "CONTROL ERRORS RETURN" kodiert, wird das Fehlerpanel nicht ge-
zeigt, aber die Verarbeitung fortgesetzt. 

2.11.2. Fehler in Dialog Manager Services 

Dialog-Service-Fehler werden von ISPF selbst behandelt. Dialog Manager meldet entweder das feh-
lerhafte Statement, oder einen entsprechenden Return Code. Auch in diesem Fall sind die Reaktio-
nen von bestimmten Umständen abhängig: 

• Ist der ISPF-Testmodus ausgeschaltet, wird die Edit-Session abgebrochen und in das PRI-
MARY-OPTION-MENU zurückgekehrt 

• Im Testmodus kann nach Anzeige des Fehlers im entsprechenden Panel durch die Eingabe 
"YES" der Ablauf fortgesetzt werden. Ist "CONTROL ERRORS RETURN" gesetzt, wird das 
ISPF-Fehlerpanel unterdrückt und ein entsprechender Return Code an die Funktion im Hin-
tergrund übergeben. 

2.11.3. Einsatz des REXX-SAY-Befehles 

Die REXX-Instruktion SAY kann bei Fehlersuche in einem Makro ein wertvolles Hilfsmittel sein. Auf 
diese Weise können simple Hinweise am Bildschirm ausgegeben werden, um zu erkennen, wie weit 
das Makro fehlerfrei gelaufen ist. Sofern sinnvoll, können an beliebigen Stellen des Makros auch die 
Inhalte verschiedener Variabler ausgegeben werden. 

Als kleine Demonstration erweitern wir das Makrobeispiel MACDEMO2 mit ein paar SAY-Befehlen. 
Auftretende Fehler können so leicht erkannt werden: 

/* MACDEMO2 *********************************/ 

ADDRESS ISREDIT "MACRO" 

DO CNT=1 TO 3 

   SAY "Wert von CNT am Schleifenanfang:" CNT 

   ADDRESS ISREDIT "FIND 'ZEILE-?'" 

   RCODE=RC 

   SAY "Returncode des FIND-Befehls:" RCODE 

   IF RCODE = 0 THEN DO 

      ADDRESS ISREDIT "CHANGE '?'" cnt 

      SAY "Ergebnis des CHANGE-Befehles:" RC 

   END 

   ELSE CNT=3 

END 

EXIT 0 

Das Zwischenspeichern der Variablen RC in einer symbolischen Variablen RCODE ist sinnvoll, da 
RC nach jedem Befehl neu gesetzt wird. 

Die Ergebnisse der SAY-Befehle erscheinen während des Ablaufes am Terminal und führen zu einer 
scheinbar schrittweisen Verarbeitung. Tritt während der Ausführung des Makros kein Fehler auf, 
erscheint nach Makroverarbeitung hinter der letzten durch SAY ausgegebenen Zeile der Hinweis für 
logisches Bildschirmende(***).  

Wert von CNT am Schleifenanfang: 1 

Returncode des FIND-Befehls: 0 

Returncode des CHANGE-Befehls: 0 

Wert von CNT am Schleifenanfang: 2 

Returncode des FIND-Befehls: 0 

Returncode des CHANGE-Befehls: 0 

Wert von CNT am Schleifenanfang: 3 

Returncode des FIND-Befehls: 0 

Returncode des CHANGE-Befehls: 0 

*** 

Nach ENTER zeigt der Editor die veränderten Daten. 

2.11.4. Einsatz des TRACE-Befehles unter REXX 

Durch den TRACE-Befehl lässt sich ein REXX-Programm sehr detailliert im Ablauf verfolgen. Die 
Möglichkeiten gehen sehr viel weiter, als dies vergleichsweise in einer CLIST über die Instruktion 
CONTROL möglich ist. Auf sämtliche Details des Trace soll hier nicht eingegangen werden, dennoch 
die wichtigsten Optionen : 

TRACE erlaubt zwei mögliche, wahlfreie  Prefix-Angaben. Kodiert werden können ! und ? mit fol-
gender Auswirkung: 

2.11.4.1. ! Syntaxprüfung der REXX-Statements ohne Ausführung 

Für jeden Befehl wird Return Code 0 angenommen. Wird im interaktiven TRACE-Modus die Be-
fehlsausführung unterdrückt, kann dies durch Eingabe einer beliebigen TRACE-Option mit einem 
führenden Rufzeichen (!) aufgehoben werden. Mehrfache !-Angaben innerhalb einer Prozedur be-
wirken immer das Gegenteil der aktuellen Situation. Generelles Ausschalten der Befehlsunterdrü-
ckung kann durch TRACE OFF oder TRACE ohne weitere Angaben erreicht werden. Diese Instruk-
tion kann auch im interaktiven TRACE-Modus während des Ablaufes abgesetzt werden. 

2.11.4.2. ? Interaktiver TRACE-Modus 

Wird im interaktiven Tracemodus gearbeitet (z.B.: TRACE ?R), setzt der Interpreter nach jeder Be-
fehlsausführung einen Haltepunkt und zeigt das Ergebnis. Hier kann jetzt eine beliebige Komman-
doeingabe erfolgen, die zum Verweilen der Prozedur in der Halteschleife führt. Wird nur <ENTER> 
gedrückt, fährt die Prozedur in ihrem Ablauf fort. Bei Eingabe eines Gleichheitszeichens wird das 
zuletzt ausgeführte Statement wiederholt. Ansonsten kann jedes beliebige Statement eingegeben 
werden, um beispielsweise die Werte von Variablen abzufragen oder zu verändern: 

say var1 

var2=35 

trace &n 

Durch Ampersand(&) vor dem Parameter im TRACE wird der interaktive Modus verlassen und ge-
mäß der angegebenen Option (hier NORMAL) weitergearbeitet. Soll die EXEC ganz verlassen wer-
den, kann dies durch EXIT erreicht werden.  

Wird im Ablauf mehrfach eine Traceoption mit ? angegeben, wird das jeweilige Gegenteil zur aktu-
ellen Situation (einschalten oder ausschalten des interaktiven Modus) erreicht. Ein generelles Aus-
schalten der interaktiven Ablaufverfolgung kann durch die Instruktion TRACE OFF oder TRACE 
ohne weitere Angaben erreicht werden. Diese Instruktion kann auch während des Ablaufs im inter-
aktiven Modus abgesetzt werden. 

Wird im interaktiven TRACE eine integere positive Zahl eingegeben, hat dies zur Folge, dass bei-
spielsweise bei TRACE 100 für die nächsten 100 Instruktionen der interaktive Modus ausgeschaltet 
wird. Danach wird normal mit der Ablaufverfolgung fortgefahren. 

Mit oder ohne Prefixangabe können die Traceoptionen in Form von Buchstabencodes gesetzt wer-
den. Die Buchstabenkürzel haben folgende Bedeutung: 

▪ A (ALL)  
Alle Befehle werden vor ihrer Ausführung gezeigt. 

▪ I (INTERMEDIATES)  
Zwischenergebnisse während der Auflösung von Ausdrücken oder Variablen werden ge-
zeigt. 

▪ N (NORMAL)  
Befehle, die nach ihrer Ausführung einen negativen Return Code melden, werden protokol-
liert (Variablenzuweisung außerhalb eines Befehles). Die Ablaufverfolgung wird ausgeschal-
tet. 

▪ R (RESULTS)  
Die Ergebnisse bei Auflösung von Variablen oder Ausdrücken, sowie die Wertzuordnung von 
Variablen durch PULL, ARG und PARSE werden gezeigt. Dies ist die häufigste Anwendung. 

▪ S (SYNTAX)  
Die gesamte Syntax wird überprüft, ohne dass eine Anweisung ausgeführt wird. Diese Option 
ist vor dem ersten Ablauf sehr zu empfehlen, da hierbei beispielsweise ein fehlendes END 
erkannt werden könnte. 

2.11.4.3. Ausgabeformat des TRACE 

Die unterschiedlichen TRACE-Optionen werden beim Debugging durch entsprechende Identifikati-
onen gekennzeichnet. Jedes gemeldete Source-Statement wird mit seiner Zeilennummer ausgege-
ben. Ist die Zeilennummer größer 99999, wird links abgeschnitten und ein Fragezeichen vorange-
stellt (z.B.: bei Zeilennummer 100487 erscheint ?00487). Alle gezeigten Sourcezeilen bekommen 
während des TRACE einen drei Byte langen Vorspann, um die Art des TRACE auszudrücken. Die-
ser kann sein: 

*-*  

Befehlszeile im Sourcecode 

+++  

TRACE-Message. Kann ein Return Code ungleich 0 eines Kommandos oder ein Syntaxfehler in 
einer Instruktion sein. 

>>>  

Resultat eines Ausdruckes (bei TRACE R), oder einer Wertzuordnung für eine Variable durch 
PARSE, oder ein zurückgegebener Wert von einer Subroutine. 

2.11.5. Edit Makro Return Codes 

Ein Edit Makro kann folgende Return Codes ausgeben. Die Codes beeinflussen die Befehlszeile 
und die Cursorposition bei der nächsten Anzeige des Edit-Panels: 

0 Normales Ende des Makros.  Die Cusorposition entspricht den Angaben im Makro. 

1 Normales Ende des Makros.  Der Cursor steht in einer leeren Befehlseingabezeile. 
Dieser Code kann benutzt werden, um den Aufruf eines weiteren Makros oder Befehles im 
direkten Anschluss zu erleichtern. 

4/8 sind reserviert und dürfen nicht gesetzt werden, sind aber gleichbedeutend mit 0 

ab 12 werden Fehler gemeldet.   
Der Cursor steht in der Befehlszeile und der Makroaufruf ist linksbündig sichtbar. Return 
Code 12 wird in Verbindung mit dem Dialog Manager Service SETMSG eingesetzt, um 
einen Bediener auf falsche oder fehlende Parameter aufmerksam zu machen. 

2.11.6. Editor Return Codes 

Jeder Makrobefehl setzt die Variable RC mit einem entsprechenden Return Code. Die Codes kön-
nen sich zwischen 0 und 20 bewegen: 

0 Normales Ende 

2, 4, 8 Hinweismeldung. Kann innerhalb des Makros abgefragt oder ignoriert werden.  

ab 12 werden Fehler gemeldet. Im Regelfall wird das Makro abnormal beendet und ein Fehlerpa-
nel mit Hinweis und der verursachenden Zeile ausgegeben. 

2.11.7. Fehlersteuerung 

In manchen Situationen kann ein Makro einen Return Code selbst behandeln (bis einschließlich 8). 
Fehler (ab Return Code 12) führen im Regelfall zum Abbruch. Sollen auch die Return Codes 12 und 
größer vom Makro behandelt werden, muss im Makro 

ADDRESS ISPEXEC "CONTROL ERRORS RETURN" 

kodiert sein. Die Standardeinstellung für Fehlersteuerung im Dialog Manager lautet: 

ADDRESS ISPEXEC "CONTROL ERRORS CANCEL" 

und führt bei Return Codes 12 oder mehr zum Abbruch. Prinzipiell kann das CONTROL-Statement 
im Makro beliebig oft mit unterschiedlichen Parametern kodiert werden, um die Fehlerbehandlung 
situationsabhängig ein- oder auszuschalten. 

2.12. Beispiele Edit Makros 

2.12.1. Meldungsausgabe über ein Makro. 

Das Beispiel zeigt, wie ein Edit Makro mit dem Anwender kommunizierten kann. SAY-Befehle in 
REXX sind in der Regel nur ein Hilfsmittel bei der Programmierung. Für die Kommunikation mit dem 
Anwender sollten in einer ISPF-Umgebung die ISPF-Messages benutzt werden. 

/* REXX * EDIT MACRO  

Macroname.: $MSG 

Zweck.....: Beispiel für Meldung aus Makro 

*/ 

address isredit 

"macro" 

zedsmsg="Kurzmlg. aus Macro" 

zedlmsg="Lange Meldung aus Macro" 

address ispexec "setmsg msg(isrz000)" 

exit 

2.12.2. Ermitteln Membergröße 

Das Macro ermittelt die Anzahl der Zeilen, welche ein Member aktuell enthält. 

/* REXX * EDIT MACRO 

Macroname.: $MSIZE 

Zweck.....: Liefert die Membergröße in Zeilen 

*/ 

address isredit 

"macro" 

"(fline) = linenum .zfirst" 

"(lline) = linenum .zlast" 

zedsmsg=lline 

zedlmsg=lline "Zeilen gezählt" 

address ispexec "setmsg msg(isrz000)" 

exit 

2.12.3. Command-Alias 

Das Setzen von Command Synonymen ist in vielen Fällen eine Arbeitserleichterung. Das Beispiel 
zeigt, wie durch Eingabe eines "S" in der Edit Command Line das SAVE-Kommando ausgeführt 
wird. 

/* REXX * EDIT MACRO 

Macroname.: $SAVE 

Zweck.....: setzt Alias S für Primary-Cmd SAVE 

*/ 

address isredit 

"macro" 

"define s alias save" 

zedsmsg="SAVE-Alias o.k." 

zedlmsg="Der Primärbefehl SAVE kann mit S" , 

        "abgekürzt werden" 

address ispexec "setmsg msg(isrz000)" 

exit 

2.12.4. Standard Profil 

Setzt die diversen Parameter in einem Dateiprofil auf gewünschte Werte. Wird das Makro als Initial-
Macro im Edit Entry Panel angegeben, erfolgt die Bearbeitung von Daten immer mit diesen Profi-
leinstellungen unabhängig vom LLQ der Datei. 

/* REXX * EDIT MACRO 

Macroname.: $STDPROF 

Zweck.....: Erzeugt Standard-Profil 

*/ 

address isredit 

"macro" 

"(lmode) = profile" 

"autolist off" 

"autonum off" 

"autosave on" 

"caps off" 

"hex off" 

"nulls on" 

"number off" 

"recovery on" 

"stats on" 

"tabs off" 

if lmode = "LOCK" then stat = "temporär verändert" 

else stat = "mit Standards gespeichert" 

zedsmsg="Profile O.K." 

zedlmsg="Profile" stat 

address ispexec "setmsg msg(isrz000)" 

exit 

2.12.5. Initialisierungsmacro 

Hier wird verschiedenes erledigt. Neben der Definition eines Standart-Profils werden auch Shortcuts 
definiert und das richtige HILITE-Kommando abgesetzt. Die Art des HILITEs wird von verschiedenen 
Dingen abhängig gemacht. Seht selbst: 

/* REXX * EDIT MAKRO */                                                          

parse source . RxTYP RxName .                                                    

address isredit                                                                  

"macro"                                                                          

if rc > 0 then call abend RxName "ist ein Init-Macro."                           

"define s alias save"                                                            

call Set_Vars                                                                    

if Set_Profile then do                                                           

   "CAPS     = OFF"                                                              

   "NULLS    = ON"                                                               

   "TABS     = OFF"                                                              

   "AUTOSAVE = ON"                                                               

   "AUTONUM  = OFF"                                                              

   "AUTOLIST = OFF"                                                              

   "STATS    = ON"                                                               

   "PACK     = OFF"                                                              

   "NOTE     = OFF"                                                              

   "(NUM)    = NUMBER"                                                           

   if NUM <> "OFF" then "UNNUM"                                                  

   select                                                                        

      when REXX  then Type = "REXX LOGIC"                                        

      when PANEL then Type = "PANEL"                                             

      when SKEL  then Type = "SKEL"                                              

      when JCL   then Type = "JCL"                                               

      otherwise       Type = "AUTO"                                              

   end                                                                           

   "HILITE" Type "PAREN CURSOR FIND"                                             

end                                                                              

exit                                                                             

Set_Vars:                                                                        

   /******************************************************************           

   * Makro-Zuweisungen                                               *           

   ******************************************************************/           

   "(DSN)      = dataset"                                                        

   "(Mem)      = member"                                                         

   "(FirstLn)  = linenum .zf"                                                    

   Empty       = (RC <> 0)                                                       

   "(LastLn)   = linenum .zl"                                                    

   "(SetNum)   = number"                                                         

   "(SetCaps)  = caps"                                                           

   /******************************************************************           

   * Lesen Screen-Buffer                                             *           

   ******************************************************************/           

   address ispexec "vget (zscreeni zscreenc)"                                    

   upper zscreeni                                                                

   /******************************************************************           

   * REXX-Zuweisungen                                                *           

   ******************************************************************/           

   LLQ         = substr(dsn,lastpos(".",dsn)+1)                                  

   Work4       = "" !! ,                       /* Unterstützte LLQs */           

                 "DATA"                      , /* Daten allgemein   */           

                 "EXEC CLIST"                , /* Scripting         */           

                 "PANELS SKELS MSGS"         , /* ISPF              */           

                 "CNTL"                      , /* Job Control       */           

                 !! ""                                                           

   Set_Profile = (wordpos(llq,work4) > 0)                                        

   /******************************************************************           

   * Typen-Erkennung für HILITE-Befehl                               *           

   ******************************************************************/           

   if ^Empty then do                                                             

      "(Line1)  = line 1"                                                        

      REXX  = left(space(line1,0),2) = "/*" & pos("REXX",Line1) > 0              

      JCL   = left(space(line1,0),2) = "//"                                      

      Panel = pos(")ATTR",zscreeni) > 0 & pos(")BODY",zscreeni) > 0              

   end                                                                           

   ELSE do                                                                       

      REXX  = LLQ = "EXEC"                                                       

      JCL   = LLQ = "CNTL"                                                       

      Panel = left(LLQ,5) = "PANEL"                                              

   end                                                                           

   Skel      = left(LLQ,4) = "SKEL"                                              

return                                                                           

                                                                                 

abend:parse arg zeldmsg;address ispexec "setmsg msg(isrz000)";exit               

                                                                                 

/* Programmbeschreibung ***********************************************          

                                                                                 

   Name....: $INITMAC                                                            

   Zweck...: Initialisierungsmakro, welches das SAVE-Command verbiegt und        

             das Dateiprofil anhängig von LLQ oder Inhalt Zeile1 setzt           

   -- Update History --------------------------------------------------          

                                                                                 

      01.2020 Nieslprim Erstellung                                               

*/ 

2.12.6. SEEK in Schleife 

Der Vorteil von SEEK liegt (im Gegensatz zu FIND) darin, dass der Zeilenstatus einer Zeile, welche 
das Suchkriterium enthält, nicht verändert wird. Wird FIND in einer ausgeblendeten Zeile fündig, 
wird die betreffende Zeile wieder sichtbar gemacht. Bei SEEK bleibt der Zeilenstatus unverändert. 

Das Beispiel führt letztlich ein CHANGE ALL Kommando durch. Zeilen, welche das Change-Krite-
rium enthalten, aber ausgeblendet sind, bleiben auch nach dem Change excluded. 

/* REXX * EDIT MACRO 

Macroname.: $CHG 

Zweck.....: SEEK in Schleife mit Zeilenstatusabfrage 

*/ 

changed=0 

lnumstr="" 

address isredit 

"macro (parm1 parm2 parm3)" 

"seek '&parm1' &parm3" 

do while rc = 0 

   "(lnum) = linenum .zcsr" 

   lnum=strip(format(lnum,6),l) 

   lnumstr=lnumstr lnum 

   "(zeilstat) = xstatus .zcsr" 

   "change '&parm1' '&parm2' &parm3" 

   changed=changed+1 

   "xstatus .zcsr = (zeilstat)" 

   "seek '&parm1' &parm3" 

end 

zedsmsg=changed 

zedlmsg=lnumstr 

address ispexec "setmsg msg(isrz000)" 

exit 

2.12.7. Zeilen ausblenden (Zeilenweise Lesen) 

Das Ergebnis könnte im Editor auch durch die Befehlsfolge X ALL;FIND ALL …;FLIP hergestellt 
werden. FLIP gab es nicht immer. Das Beispiel benutzt dazu REXX-Logik. 

/* REXX * EDIT MACRO 

Macroname.: $SHOW1 

Zweck.....: Ausblenden aller Zeilen, die den 

            Suchbegriff nicht enthalten. 

            Boundary-Angaben werden berücksichtigt 

*/ 

found=0 

address isredit 

"macro (parm1 parm2 parm3)" 

"(links,rechts) = bounds" 

"(recl) = lrecl" 

if datatype(parm2)="CHAR" ! parm2>rechts then parm2=links 

if datatype(parm3)="CHAR" ! parm3 < parm2 ! , 

   parm3 > recl then parm3 = rechts 

"(first) = linenum .zfirst" 

"(last) = linenum .zlast" 

"exclude all" 

do pointer = first to last 

   "(linestr) = line" pointer 

   spalte = pos(parm1,linestr,parm2) 

   if spalte > 0 & spalte <= parm3 then do 

      "xstatus" pointer "= NX" 

      found=found+1 

   end 

end 

zedsmsg=found 

zedlmsg="Positives Suchergebnis in" found "Zeile(n)" 

address ispexec "setmsg msg(isrz000)" 

exit 

2.12.8. Zeilen ausblenden(EXCLUDE/FIND) 

Das gleiche Beispiel unter Verwendung des FIND-Befehles macht die Schleifenprogrammierung und 
Erkennung auf gefunden innerhalb der Programmlogik überflüssig. Die Befehle sprechen für sich 
selbst. 

/* REXX * EDIT MACRO  

Macroname.: $SHOW2 

Zweck.....: Exclude aller Zeilen, die Suchbegriff nicht enthalten 

            Boundary-Angaben werden berücksichtigt 

            Matching über FIND-Befehl 

*/ 

address isredit 

"macro (parm1 parm2 parm3)" 

"(links,rechts) = bounds" 

"(recl) = lrecl" 

if datatype(parm2)="CHAR" ! parm2 > rechts then parm2 = links 

if datatype(parm3)="CHAR" ! parm3 < parm2 ! , 

   parm3 > recl then parm3 = rechts 

"(first) = linenum .zfirst" 

"(last) = linenum .zlast" 

"exclude all" 

"find all" parm1 parm2 parm3 

"(found) = find_counts" 

"flip" 

zedsmsg=found 

zedlmsg=parm1 "in" found "Zeilen gefunden" 

address ispexec "setmsg msg(isrz000)" 

exit 

2.12.9. Belegung PF-Taste 

Hier wird die PF13-Taste mit einem Macro belegt. Die Aktuelle Belegung der PF13-Taste wird im 
Profile Pool abgelegt und kann zu einem späteren Zeitpunkt wieder hergestellt werden. 

/* REXX * EDIT MACRO 

Macroname.: $PF13BOX 

Zweck.....: Belegt PF13 mit Aufruf für Kommentarbox  

*/ 

address isredit 

"macro" 

address ispexec "vget (zpf13) profile" 

if translate(zpf13) ^= "$CMBOX" then do 

   pf13alt=zpf13 

   zpf13="$CMBOX" 

   address ispexec "vput (pf13alt zpf13) profile" 

   zedsmsg="PF13 = BOX" 

   zedlmsg="PF13-Taste erzeugt eine Kommentarbox" 

   address ispexec "setmsg msg(isrz000)" 

end 

exit 

 

/* REXX * EDIT MACRO $PF13RES 

Zweck.....: Setzt PF13 wieder auf Benutzerangabe 

*/ 

address isredit 

"macro" 

address ispexec "vget (zpf13 pf13alt) profile" 

if zpf13 = "$CMBOX" then do 

zpf13=pf13alt 

address ispexec "vput (zpf13) profile" 

address ispexec "verase (pf13alt) profile" 

zedsmsg="PF13 = "zpf13 

zedlmsg="PF13-Taste wieder auf" zpf13 "gesetzt" 

address ispexec "setmsg msg(isrz000)" 

end 

exit 

2.12.10. Output Trapping 

Erlaubt die Eingabe eines Befehles, dessen Output aufgefangen, in eine Dateigeschrieben und per 
VIEW angezeigt wird. Auf diese Weise kann beispielsweise die HELP-Funktion sehr komfortabel 
benutzt werden und in der Anzeige lässt sich blättern und suchen. 

/* REXX 

   TRAP Output des ARGs und Anzeige über Tempfile im VIEW 

*/ 

arg cmd 

address isredit "macro (parm)" 

if rc = 0 then cmd = parm 

call outtrap info. 

interpret cmd 

call outtrap "OFF" 

  

address ispexec "vget (ztempf ztempn)" 

"alloc dd(trap) da('"ztempf"') shr reuse" 

"execio * diskw trap (stem info. finis" 

address ispexec 

"LMINIT DDNAME(&ZTEMPN) DATAID(TRAP) ENQ(SHR)" 

"VIEW   DATAID(&TRAP)" 

"LMFREE DATAID(&TRAP)" 

Address tso "free dd(trap)" 

exit 

2.12.11. Autovervollständiger.  

Bei der Erstellung von REXX-Programmen werden häufig Select-Gebilde kodiert. Diese haben in 
der Regel mindestens 5 Zeilen (11, wenn die THEN-Zweige mit DO/END-versehen sind). Da SEL-
ECT der einzige Befehl ist, der mit SE beginnt, wäre nach 2 Tastenanschlägen schon klar, was der 
Anwender will. Also: SE tippen – das Macro per PF-Taste aufrufen – es untersucht den Inhalt der 
aktuellen Zeile und schreibt das SELECT-Gebilde spaltensensitiv fertig. Das Beispielmacro kann 
auch für Vervollständigung von JCL benutzt werden und ist beliebig erweiterbar. 

/* REXX * EDIT MACRO ************************************ 

  

   Name.: #TF (TippFertig) 

   Zweck: Erkennt Instruktion in der Cursor-Zeile und 

          ergänzt den Befehl 

  

Erstellt: 01.94 Wittemann 

Geändert: 07.08 Wittemann 

*/ 

address isredit 

"macro" 

"(dsn) = DATASET" 

"(csr,col) = CURSOR" 

"(CMD) = line .zcsr" 

upper cmd 

parse value reverse(dsn) with llq "." 

llq = reverse(llq) 

rexx = ("EXEC" = llq) 

jcl  = ("CNTL" = llq) 

  

select 

   when rexx then do 

      start=pos(strip(cmd),cmd) 

      cmdstrip=strip(cmd,b) 

      blanks=copies(' ',start-1) 

   end 

   when jcl then do 

      cmd = strip(cmd) 

      parse var cmd . 3 stepname , 

          "EXEC" . "=" ?pgm "," jclparm 

      if strip(stepname) = "" then do 

         cmd = "//"left(?pgm,8) "EXEC PGM="?pgm 

         if jclparm <> "" then cmd = cmd","jclparm 

      end 

   end 

   otherwise do 

      zedsmsg = "nur EXEC und CNTL" 

      zedlmsg = "Keine Unterstützung für" llq". Macro" , 

         "läuft nur in EXEC (REXX) oder CNTL (JCL)" 

      address ispexec "setmsg msg(isrz000)" 

      exit 

   end 

end 

  

select 

   when ?pgm = "ICEMAN"            then label="DFSORT" 

   when ?pgm = "DFSORT"            then label="DFSORT" 

   when ?pgm = "SORT"              then label="DFSORT" 

   when ?pgm = "IEBGENER"          then label="IEBGENER" 

   when ?pgm = "PDSFAST"           then label="IEBCOPY" 

   when ?pgm = "IEBCOPY"           then label="IEBCOPY" 

   when ?pgm = "IEBCOPYA"          then label="IEBCOPY" 

   when pos(cmdstrip,"IF")     > 0 then label="IF" 

   when pos(cmdstrip,"SELECT") > 0 then label="SELECT" 

   when word(cmdstrip,1) = "DO"    then label="DO" 

   otherwise do 

      if REXX then do 

         zedsmsg="Unbekannter Befehl" 

         zedlmsg="Unterstützt werden IF, SELECT und DO" 

      end 

      else do 

         zedsmsg="Unbekanntes Programm" 

         zedlmsg="Möglich: DFSORT, IEBCOPY, IEBGENER" 

      end 

      address ispexec "setmsg msg(isrz000)" 

      exit 

   end 

end 

interpret "CALL" label 

"line .zcsr = '"zeile.0"'" 

do cnt = 1 while zeile.cnt <> "ZEILE."cnt 

   parse var zeile.cnt zeile.cnt "²" linetype 

   if linetype = "" then linetype = "DATALINE" 

   "line_after .zcsr = &linetype '"zeile.cnt"'" 

end 

exit 

  

if: 

   zeile.0=blanks"if BEDINGUNG then do" 

   zeile.5=blanks"   nop" 

   zeile.4=blanks"end" 

   zeile.3=blanks"else do" 

   zeile.2=blanks"   nop" 

   zeile.1=blanks"end" 

return 

  

select: 

   zeile.0 =blanks"select" 

   zeile.10=blanks"   when BEDINGUNG then do" 

   zeile.9 =blanks"      nop" 

   zeile.8 =blanks"   end" 

   zeile.7 =blanks"   when BEDINGUNG then do" 

   zeile.6 =blanks"      nop" 

   zeile.5 =blanks"   end" 

   zeile.4 =blanks"   otherwise do" 

   zeile.3 =blanks"      nop" 

   zeile.2 =blanks"   end" 

   zeile.1 =blanks"end" 

return 

do: 

   parm=word(cmdstrip,2) 

   select 

      when parm = ""      then loopkopf="do" 

      when pos('=',cmd)>0 then loopkopf="do VAR = ? to ?" 

      when pos(parm,"WHILE")>0 then loopkopf="do while ?" 

      when pos(parm,"UNTIL")>0 then loopkopf="do until ?" 

      otherwise loopkopf=cmd 

   end 

   zeile.0=blanks!!loopkopf 

   zeile.2=blanks"   nop" 

   zeile.1=blanks"end" 

return 

  

dfsort: 

   zeile.0 = cmd 

   zeile.19= "//SORTIN   DD DISP=SHR,DSN=???" 

   zeile.18= "//"copies("*",68)"²NOTELINE" 

   zeile.17= "//* SORTOUT existiert²MSGLINE" 

   zeile.16= "//SORTOUT  DD DISP=OLD,DSN=???²NOTELINE" 

   zeile.15= "//"copies("*",68)"²NOTELINE" 

   zeile.14= "//* SORTOUT existiert nicht²MSGLINE" 

   zeile.13= "//SORTOUT  DD DSN=???,²NOTELINE" 

   zeile.12= "//       DISP=(NEW,CATLG),²NOTELINE" 

   zeile.11= "//       UNIT=(unit),²NOTELINE" 

   zeile.10= "//       SPACE=(sp,(pri,sec,dir))²NOTELINE" 

   zeile.9 = "//       LIKE=dateiname²NOTELINE" 

   zeile.8 = "//SORTWK01 DD UNIT=VIO,SPACE=(CYL,(10,10))" 

   zeile.7 = "//SORTWK02 DD UNIT=VIO,SPACE=(CYL,(10,10))" 

   zeile.6 = "//SORTWK03 DD UNIT=VIO,SPACE=(CYL,(10,10))" 

   zeile.5 = "//SORTWK04 DD UNIT=VIO,SPACE=(CYL,(10,10))" 

   zeile.4 = "//SYSOUT   DD SYSOUT=*" 

   zeile.3 = "//SYSIN    DD *" 

   zeile.2 = "  sort_sysin_angaben²NOTELINE" 

   zeile.1 = "/*" 

return 

  

iebgener: 

   zeile.0 = cmd 

   zeile.4 = "//SYSUT1   DD DISP=SHR,DSN=???" 

   zeile.3 = "//SYSUT2   DD SYSOUT=*" 

   zeile.2 = "//SYSPRINT DD SYSOUT=*" 

   zeile.1 = "//SYSIN    DD DUMMY" 

return 

  

iebcopy: 

   zeile.0 = cmd 

   zeile.17= "//SYSUT1   DD DISP=SHR,DSN=???" 

   zeile.16= "//"copies("*",68)"²NOTELINE" 

   zeile.15= "//* SYSUT2 existiert²MSGLINE" 

   zeile.14= "//SYSUT2   DD DISP=OLD,DSN=???²NOTELINE" 

   zeile.13= "//"copies("*",68)"²NOTELINE" 

   zeile.12= "//* SYSUT2 existiert nicht²MSGLINE" 

   zeile.11= "//SYSUT2   DD DSN=???²NOTELINE" 

   zeile.10= "//      DISP=(NEW,CATLG),²NOTELINE" 

   zeile.9 = "//      UNIT=(unit),²NOTELINE" 

   zeile.8 = "//      SPACE=(spc,(pri,sec,dir))²NOTELINE" 

   zeile.7 = "//      LIKE=dateiname²NOTELINE" 

   zeile.6 = "//SYSUT3   DD UNIT=VIO,SPACE=(CYL,(10,10))" 

   zeile.5 = "//SYSUT4   DD UNIT=VIO,SPACE=(CYL,(10,10))" 

   zeile.4 = "//SYSPRINT DD SYSOUT=*" 

   zeile.3 = "//SYSIN    DD *" 

   zeile.2 = "  COPY INDD=((SYSUT1,R)),OUTDD=SYSUT2" 

   zeile.1 = "/*" 

return 

 

2.12.12. Nachricht senden 

Hier eine Art von Informationsübergabe an einen anderen Benutzer, wenn Sametime nicht verfügbar 
ist. Tipp: durch eine kleine Änderung könnte das Macro die Zeilen, welche es gesendet hat, leer 
machen. Sieht man davon ab, dass es mit zumutbarem Aufwand nicht möglich ist, Emoticons mit zu 
schicken, käme man damit nahe an die Sametime-Funktionalität. 

/* REXX * SCHICKS  

   Sendet Spalten 1 bis 70 aller Zeilen,  

   die im Zugiff des Editors liegen an den  

   angegebenen Benutzer 

*/ 

address isredit 

"macro (pseudo,user)" 

call msg "OFF" 

call msg "ON" 

upper PSEUDO USER 

if wordpos(PSEUDO,"ZU ZUM AN NACH FUER") = 0 then USER = PSEUDO 

if USER = "" then user = userid() 

"(last) = linenum .ZL" 

do I = 1 to LAST 

   "(xstat) = XSTATUS" i 

   if xstat = "X" then iterate 

   "(msg) = LINE" I 

   msg = translate(left(msg,70),'"',"'") 

   address tso "send '"msg"' user("USER")" 

end 

exit 

2.12.13. REXXe speichern und starten 

Für die Ausführung von JCL steht im Editor der SUBmit-Befehl zur Verfügung, welcher die Daten, 
die der Editor im Zugriff hält, an den Internal Reader übergibt. Aus diesem Grund muss vor einer 
JCL-Verarbeitung auch kein SAVE erfolgen. 

REXX-Programme dagegen müssen vor Ausführung erst gespeichert werden (sonst läuft die alte 
Version), da der Interpreter das Programm immer von Platte liest. Die Art des Aufrufes ist davon 
abhängig, ob die Datei, in welcher das Programm liegt, unter SYSPROC/SYSEXEC verkettet ist. 

Das Macro RUN übernimmt das Speichern, ermittelt den Datei- und Membernamen des Programms 
und führt mit diesen Informationen im Hintergrund einen TSO-EXEC-Befehl aus. Dabei wird syntak-
tisch unterschieden, ob es sich um CLIST oder REXX handelt. 

/* REXX * EDIT MACRO * RUN 

   Führt intern einen SAVE-Befehl aus und 

   Startet das Programm über den TSO-Befehl EXEC 

   Unterstützt CLISTen und REXXe 

*/ 

address isredit 

"macro (params)" 

"(dsn) = dataset" 

if pos(".EXEC",DSN) + pos(".CLIST",DSN) > 0 then do 

   "(mem) = member" 

   "save" 

   if pos(".EXEC",DSN) > 0 then params = params"' EXEC" 

   address tso "ex '"dsn"("mem")' '"params 

end 

else do 

   ZEDSMSG = "Falscher Bibliothekstyp" 

   ZEDLMSG = DSN "ist keine REXX oder CLIST Bibliothek" 

   address ispexec "setmsg msg(ISRZ000)" 

end 

exit 

2.12.14. Starten von REXXen, CLISTen, JCL und mehr 

/* REXX 

 

   Programmbeschreibung und Update History am Ende 

 

*/ 

parse source . . RxName . 

address isredit 

"macro (params)" 

 

if RC <> 0 then do 

   zedlmsg = RxName" ist ein Edit Macro." 

   address ispexec "setmsg msg(ISRZ000)" 

   exit 

end 

address ispexec "control errors return" 

 

call Set_Vars 

call What_Shall_We_Do 

 

if REXX ! CLIST ! NETCL ! SHELL then do 

   "save" 

   if rc > 4 then do 

      zedlmsg = "Keine Verarbeitung, SAVE endete mit RC:" rc 

      address ispexec "setmsg msg(isrz001)" 

      exit 

   end 

end 

address tso 

select 

   when JCL   then address isredit "SubMit &Params" 

   when REXX  then "EX '"dsn"("mem")' '"params"' EXEC" 

   when CLIST then "EX '"dsn"("mem")' '"params"'" 

   when NetCl then "EX '"dsn"("mem")' '"params"'" 

   when SHELL then call ShellScript 

   otherwise  ZEdLMsg = RxName , 

              "unterstützt REXX, CLIST, JCL und Shell Script." 

End 

if strip(ZEdSMsg ZEdLMsg)<>"" then address ispexec "setmsg msg(&MsgMbr)" 

exit 

 

ShellScript: 

   MyRC = bpxwunix(dsn params, , stdout., stderr.) 

   say center(" STDOUT: ",linesize()-1,"-") 

   do i=1 to stdout.0 

     say stdout.i 

   end 

   say center(" STDERR: ",linesize()-1,"-") 

   do i=1 to stderr.0 

     say stderr.i 

   end 

   ZEdSMsg = "RC =" myrc 

   if MyRC <> 0 then MsgMbr = "ISRZ000" 

return 

 

Set_Vars: 

   parse value 0 with 1 JCL 1 REXX 1 CLIST 1 NetCL 1 SHELL 

   MsgMbr   = "ISRZ000" 

   ZedSMsg  = 

   ZedLMsg  = 

   "(dsn)   = dataset" 

   "(mem)   = member" 

   "(Line1) = line 1" 

   if pos(".",dsn) > 0 then LLQ = substr(DSN,lastpos(".",dsn)+1) 

   RxClst_Q = "CLIST DSICLD EXEC REXX SISPEXEC SBPXEXEC SISPCLIB SRC" 

   JCL_Q    = "CNTL" 

Return 

 

what_shall_we_do: 

   /****************************************************************** 

   * Prüfung auf JCL                                                 * 

   ******************************************************************/ 

   Select 

      when strip(dsn mem) = ""                              then JCL = 1 

      when left(Line1,2) = "//" & wordpos("JOB" ,Line1) > 0 then JCL = 1 

      when left(Line1,2) = "//" & wordpos("EXEC",Line1) > 0 then JCL = 1 

      otherwise nop 

   end 

   if JCL then return 

 

   /****************************************************************** 

   * Prüfung auf Shell Script                                        * 

   ******************************************************************/ 

   Shell = (left(dsn, 1)right(dsn, 3) == "/.sh") 

   if Shell then return 

 

   /****************************************************************** 

   * Prüfung auf REXX, CLIST oder NetCl                              * 

   ******************************************************************/ 

   if wordpos(LLQ,RxClst_Q) > 0 then do 

   "(Line1) = Line 1" 

   parse upper value strip(Line1) with 1 First2 3 Rest 1 Word1 Word2 

   select 

      when First2 = "/*" & pos("REXX",Rest) = 0 then REXX  = 1 

      When Word1  = "PROC"                      then CLIST = 1 

      When wordpos("CLIST",Word1 Word2) > 0     then NetCl = 1 

      otherwise                                      REXX  = 1 

   end 

end 

return 

 

/* Programmbeschreibung ********************************************** 

 

   Edit Makro für die Ausfühlrung von: 

   - REXX 

   - CLIST 

   - NetView CLIST (alte Formate) 

   - Shell Scripts 

   - JCL 

 

   Außer bei JCL wird immer erst ein SAVE-Befehl abgesetzt. 

 

-- Update History ---------------------------------------------------- 

 

11.11.2016 Wittemann 

      ° RC 4 bei SAVE ist auch OK 

 

*/ 

2.12.15. Ermittelt Account-Info und trägt sie gegebenenfalls ein 

Das Elend mit der Account-Information. Oftmals ist die richtige Account-Info nicht bekannt. Da sie 
bei JCL-Verarbeitung aber immer häufiger geprüft wird, darf sie zum einen nicht fehlen und muss 
zum anderen auch noch richtig sein. 

Das Makro ermittelt die Account-Information, mit welcher der LOGON gelaufen ist und fügt sie ent-
weder in die getippte JCL ein oder schreibt sie als ShortMessage an den Bildschirm. 

/* REXX * Edit Macro 

  

   JACC   Ermittelt Account-Info aus Logon (ZACCTNUM). 

          Sucht die Zeichenfolge "JACC". 

          Wird diese gefunden, und beginnt die jeweilige 

          Zeile mit // und enthält " JOB ", 

          wird sie durch ZACCTNUM ersetzt. 

*/ 

parse source . rxTYP rxNAME . 

address isredit 

"macro" 

if rc > 0 then do 

   zedlmsg = rxNAME "ist ein Edit Macro" 

   address ispexec "setmsg msg(isrz000)" 

   exit 

end 

address ispexec "vget zacctnum" 

done = 0 

"up max" 

"seek 'JACC'" 

do while rc = 0 

   "(row,col)  = CURSOR" 

   "(linedata) = line &row" 

   if left(linedata,2) = "//" & , 

      pos(" JOB ",linedata) > 0 then do 

      done = 1 

      parse upper var linedata _1 "JACC" _2 

      linedata = _1 !! zacctnum !! _2 

      "line &row = '&linedata'" 

      "line_after &row = NOTELINE '*** JACC" , 

         "durch ZACCTNUM ersetzt ***'" 

      leave 

   end 

   "seek 'JACC'" 

end 

if done then zedsmsg = "JACC changed" 

else         zedsmsg = zacctnum 

address ispexec "setmsg msg(isrz000)" 

exit 

2.12.16. Erkennt Dateinamen und zeigt diese mit View 

Wird der Cursor im View oder Edit auf einen Dateinamen gestellt, ermittelt das Macro diesen und 
zeigt die entsprechende Datei an. 

/* REXX * EDIT MACRO 

   Ermittelt Dateinamen an Cursor-Position und öffnet 

   das Dataset mit VIEW oder EDIT 

   Kann auch außerhalb von Edit/View als TSO-Command benutzt werden. 

*/ 

address ispexec "control errors return" 

address isredit 

"macro (parm)" 

If rc = 0 then do 

   "(ze,sp) = CURSOR" 

   if ze = 0 ! sp = 0 then do 

      zedlmsg = "Cursor muss auf einem Dateinamen stehen" 

      address ispexec "setmsg msg(isrz000)" 

      exit 

   end 

   "(dataline) = line &ze" 

   upper parm dataline 

end 

else do 

   arg parm 

   address ispexec "vget (zscreeni zscreenc) shared" 

   dataline = translate(zscreeni) 

   sp = zscreenc+1 

end 

if left(parm,1) <> "E" then parm = "V" 

dataline = strip(translate(dataline," ","),='"""),"L") 

blank_pos = pos(" ",dataline,sp) 

parse value reverse(left(dataline,blank_pos-1)) with dsn . 

dsn = reverse(dsn) 

if left(dsn,1) = "D" & pos("(",substr(dsn,3,2)) > 0 then do 

   parse var dsn . "(" dsn 

end 

if left(dsn,1) = "(" then dsn = substr(dsn,2,) 

parse var dsn dsn "(" member ")" . 

address ispexec 

zedlmsg = 

if sysdsn(dsn) <> "OK" then do 

   dsn = "'"dsn"'" 

   if sysdsn(dsn) <> "OK" then do 

      zedlmsg = dsn "nicht gefunden" 

      "setmsg msg(isrz000)" 

      exit 

   end 

end 

if member <> "" then do 

   if right(dsn,1) <> "'" then dsn = dsn"("member")" 

   else dsn = left(dsn,length(dsn)-1)"("member")'" 

end 

if parm = "E" then "edit dataset(&dsn)" 

else               "view dataset(&dsn)" 

exit 

3. LM-Services 

Library-Management-Services sind Bestandteil des ISPF-Werkzeugkastens. Die Services werden in 
der Praxis benutzt, um in einer Anwendung Funktionalität zu liefern, die in Verbindung mit sequen-
tiellen oder untergliederten Dateien (PS oder PO) gebraucht werden. 

Hierzu gehören Dinge wie beispielsweise  

▪ die Verarbeitung von PO-Directorys 

▪ MOVE oder COPY von Dateien oder Membern 

▪ Lesen oder Schreiben in PS- oder PO-Dateien.   
Dies könnte grundsätzlich auch in diversen Programmiersprachen erledigt werden (Cobol, 
Clist, REXX), mit den LM-Services sind aber auch Änderungen in den Memberstatistiken und 
dergleichen möglich  

▪ Umbenennen oder löschen von Dateien oder PO-Membern.  
Hier könnte grundsätzlich auch ein TSO-Kommando benutzt werden. Um ein Member aus 
einer PO-Datei mittels TSO-Delete-Befehles zu löschen, muss die betreffende Anwendung 
allerdings die exklusiven Rechte an der Datei haben. Dies ist in der Praxis oft sehr proble-
matisch. Die spezielle Form der Zuweisung eines Members mit SHARED-EXCLUSIVE ist 
nur mit den LM-Services möglich. 

Die wesentlichen Funktionen bestehen im Lesen und Schreiben, wobei mit den entsprechenden 
Serviceroutinen LMGET und LMPUT eine weitergehende Verarbeitung möglich ist, als beispiels-
weise mit EXECIO in REXX. Per LMGET lassen sich beispielsweise auch Lademodule lesen und 
Dinge wie der Timestamp aus dem Compile oder das Linkdatum ermitteln. 

Eine Vielzahl von Member-Funktionen ist über die LM-Services ebenfalls sehr viel besser durchführ-
bar, als beispielsweise über vergleichbare TSO-Befehle. So ist es auch mit dem TSO-Befehl DE-
LETE möglich, ein Member aus einem PO-Dataset zu löschen. Allerdings wird das gesamte Dataset 
vom Delete-Befehl mit einer Disposition OLD angesprochen. Ist die Datei aber nicht exklusiv zu 
bekommen, wird das Löschen nicht durchgeführt. Die LM-Services erlauben in solchen Fällen eine 
Shared-Exclusive-Belegung. Das heißt: Das Dataset wird mit SHR belegt, das Member mit OLD. 
Auf diese Weise kann ein Löschen/Umbenennen eines Members auch dann erfolgen, wenn das 
Dataset nicht exklusiv zur Verfügung steht. Dass man in Verbindung mit HSM Dinge tun kann, die 
man eigentlich nicht tun soll (darf), wollen wir hier außer Acht lassen. 

Ebenso von Vorteil sind die dateibezogenen Services. Benötigt man beispielsweise eine Dateiliste, 
wie man sie unter ISPF 3.4 mit diversen Wildcards ermitteln würde, kann man hierfür den LMDLIST-
Service aufrufen. Die Wildcards Stern(*) und Prozent(%) können im gleichen Umfang benutzt wer-
den wie unter 3.4. 

Die meisten PO-spezifischen LM-Funktionen müssen in die Befehlsfolge LMINIT-LMOPEN-Verar-
beitung-LMCLOSE eingebettet werden. 

BROWSE, VIEW und EDIT gehören ebenfalls zu den LM-Services und können gerade in Verbin-
dung mit REXX sehr komfortabel benutzt werden, um im Verlauf einer Verarbeitung mal eine Datei 
anzuzeigen, eine generierte JCL vorzulegen, die der Anwender noch editieren kann, oder eine Datei 
im Editor verdeckt aufzurufen und die Daten über ein Edit Macro zu manipulieren. 

3.1. Beispiele für LM-Services 

Die folgenden Seiten zeigen eine Auswahl von Beispielen mit REXX unter Einbeziehung der diver-
sen Services. Die Beispiele sind weitestgehend selbsterklärend. Auf die Details wird im Kapitel der 
Syntaxbeschreibung eingegangen. 

3.1.1. Lesen mit LMGET 

In den meisten Fällen wird man zum Lesen von Dateien mit REXX den TSO-Befehl EXECIO benut-
zen. Dennoch: nicht alles ist mit EXECIO möglich. Sollen beispielsweise Compile- oder Link-Infor-
mationen aus einem Lademodul gelesen werden, ist die Stunde der LM-Services gekommen. Im 
Anschluss ein simples Beispiel, welches die prinzipielle Verarbeitung von Daten mit LMGET veran-
schaulicht. 

/* REXX  

PGM-Name..: #LM1 

Zweck.....: Liest die Daten des Members ADR1 aus 

            der Datei FB132.DATA und zeigt sie an 

*/ 

arg test . 

if test = "T" then trace "?r" 

address ispexec 

"lminit dataid(did) dataset(fb132.data) enq(shr)" 

"lmopen dataid(&did) option(input)" 

"lmmfind dataid(&did) member(adr1)" 

"lmget dataid(&did) mode(invar)" , 

   "dataloc(rec) datalen(l1) maxlen(132)" 

do while rc = 0 

   say strip(rec,"T") 

   "lmget dataid(&did) mode(invar)" , 

      "dataloc(rec) datalen(l1) maxlen(132)" 

end 

"lmclose dataid(&did)" 

"lmfree dataid(&did)" 

Exit 

3.1.2. Lesen mit LMGET zur Ermittlung des Cobol-Timestamps 

Diese Prüfung erlaubt eindeutig festzustellen, ob es sich bei gleichnamigen Programmen in unter-
schiedlichen Bibliotheken wirklich um das gleiche Lademodul handelt. 

/* REXX * LMGET auf Lademodul */ 

arg lib,pgm 

loadinfo = "?" 

address ispexec  

"LMINIT DATASET('&LIB') DATAID(§DIDLIB) ENQ(SHR)" 

"LMOPEN DATAID(&§DIDLIB) OPTION(INPUT)" 

"LMMFIND DATAID(&§DIDLIB) MEMBER(&PGM)" 

do $$ = 1 

   "LMGET DATAID(&§DIDLIB) MODE(INVAR)" , 

      "DATALOC(DATA) DATALEN(LRECL) MAXLEN(32760)" 

   if rc <> 0  then leave  

   do while pos('47F0F0'X,data) > 0 

      hexpos = pos('47F0F0'X,data) 

      data = substr(data,hexpos+3) 

      if hexpos > 0 then do 

         ceepos = pos("CEE",data,hexpos) 

         if ceepos > 0 then do 

            timst = substr(data,ceepos+99,20) 

            if datatype(timst) = "NUM" then do 

               parse var timst 1 j 5 m 7 t , 

                  9 st 11 mi 13 se 15 mil  

               loadinfo =space(t"."m"."j !! , 

                  st":"mi":"se"."mil,1) 

               leave $$ 

            end 

         end 

      end 

   end 

end 

"LMCLOSE DATAID(&§DIDLIB)" 

"LMFREE  DATAID(&§DIDLIB)" 

return loadinfo  

3.1.3. Schreiben mit LMPUT 

Das Verfahren ist etwas aufwändiger als mit EXECIO bietet aber eine Reihe von Vorteilen. Einer ist 
beispielsweise des Setzen der Memberstatistiken für den Nachweis, wer wann das Member erzeugt 
hat. 

Dies könnte auch mit einem Edit Makro bewerkstelligt werden 

Das gleiche Ergebnis wäre mit dem LMMSTAT-Service im Anschluss an EXECIO erreichbar – nicht 
weniger Aufwand. 

Viele Wege führen ans Ziel. 

/* REXX * Schreiben Member mit LMPUT / LMMREP */ 

/* 

Unterstellung: 

Hier wird ein STEM CSVLLA. aufgebaut. Dieser soll als Member 

CSVLLAZZ in die Parmlib für einen LLA-Update weggeschrieben 

werden. Der Name der Parmlib liegt in der Variablen PARMLIB 

*/ 

address ispexec  

"lminit dataid(LLA) dataset('&parmlib') enq(shrw)" 

if rc ^= 0 then call abend "RC:"rc "bei LMINIT" parmlib  

"lmopen dataid(&LLA) option(output) org(po)" 

if rc ^= 0 then call abend "RC:"rc "bei LMOPEN" parmlib  

do §1 = 1 while csvlla.§1 <> "CSVLLA."§1 

   llasatz = csvlla.§1 

   "lmput dataid(&LLA) mode(invar) dataloc(LLASATZ) datalen(80)" 

end 

"lmmrep dataid(&LLA) member(CSVLLAZZ)" 

if rc > 8 then call abend "RC:"rc "bei LMMREP" parmlib "CSVLLAZZ" 

/* 

Für das Schreiben der Memberstatistiken müssen natürlich die 

Entsprechenden Informationen wie LastUpdate, Size, etc.) erst 

Ermittelt werden. 

*/ 

"lmmstats dataid(&LLA) member(CSVLLAZZ) ..." 

"lmclose dataid(&LLA)" 

"lmfree dataid(&LLA)" 

exit  

abend: procedure  

   arg zedlmsg  

   "setmsg msg(isrz000)" 

exit  

3.1.4. Auslesen der Memberstatistiken 

/* REXX 

trace ?r 

*/ 

parse source . rxTYP rxNAME . 

if rxTYP = "FUNCTION" then arg dsn,member,dummy 

else                       arg dsn member . 

call msg "OFF" 

address ispexec 

"control errors return" 

"LMINIT DATAID(DID) DATASET('&DSN') enq(SHR)" 

"LMOPEN DATAID(&DID) OPTION(INPUT)" 

"LMMFIND DATAID(&DID) MEMBER(&MEMBER) STATS(YES)" 

"LMCLOSE  DATAID(&DID)" 

"LMfree  DATAID(&DID)" 

 

if rxTYP <> "COMMAND" then return  "" , 

   "ZLINORC  = '"zlinorc"';" ,  /* Init Records  */ 

   "ZLCNORC  = '"zlcnorc"';" ,  /* Curr. Records */ 

   "ZLMNORC  = '"zlmnorc"';" ,  /* Mod. Records  */ 

   "ZLC4DATE = '"zlc4date"';" , /* Date Creation */ 

   "ZLM4DATE = '"zlm4date"';" , /* Date Modified */ 

   "ZLMTIME  = '"zlmtime"';" ,  /* Time Modified */ 

   "ZLUSER   = '"zluser"';" ,   /* user Modified */ 

   "ZLVERS   = '"zlvers"';" ,   /* Version       */ 

    "ZLMOD    = '"zlmod"';" ,   /* Mod. Level    */ 

    !! "" 

 

say "ZLINORC  = "left(zlinorc,10)   "Init Num Recs" 

say "ZLCNORC  = "left(zlcnorc,10)   "Curr Num Recs" 

say "ZLMNORC = "left(zlmnorc,10)    "Mod. Num Recs" 

say "ZLC4DATE = "left(zlc4date,10)  "Creation Date" 

say "ZLM4DATE = "left(zlm4date,10)  "Date Modified" 

say "ZLMTIME  = "left(zlmtime,10)   "Time Modified" 

say "ZLUSER   = "left(zluser,10)    "user Modified" 

say "ZLVERS   = "left(zlvers,10)    "Version" 

say "ZLMOD    = "left(zlmod,10)     "Modif. Level" 

exit 0 

3.1.5. Löschen PO-Member 

Generell kann zum Umbenennen eines PO-Members auch das TSO-Kommando DELETE benutzt 
werden. Manko ist, dass das gesamte Dataset zum Zeitpunkt des Delete exklusiv belegt wird.  

Der LM-Service LMMDEL benötigt das Dataset zum Zeitpunkt des Renames nur SHR, belegt ledig-
lich das Member exklusiv. 

/* REXX 

PGM-Name..: DELMEM 

Zweck.....: Löscht ein Member aus einer PO-Datei 

*/ 

arg datei member . 

address ispexec 

if member="" then do 

   zedsmsg="Member fehlt" 

   zedlmsg="Name des zu löschenden Members fehlt!!" 

   "setmsg msg(isrz000)" 

   exit 

end 

"lminit dataid(did) dataset(&datei) enq(shrw)" 

"lmopen dataid(&did) option(output)" 

"lmmdel dataid(&did) member(&member)" 

if rc = 0 then do 

   zedsmsg="DELETE o.k." 

   zedlmsg=member "wurde aus Datei gelöscht" 

end 

else do 

   zedsmsg="Fehler" 

   zedlmsg=member "konnte nicht gelöscht werden" 

end 

"setmsg msg(isrz000)" 

"lmclose dataid(&did)" 

"lmfree dataid(&did)" 

exit 

3.1.6. Rename Member 

Generell kann zum Umbenennen eines PO-Members auch das TSO-Kommando RENAME benutzt 
werden. Manko ist, dass das gesamte Dataset zum Zeitpunkt des Rename exklusiv belegt wird. Der 
LM-Service LMMREN benötigt das Dataset zum Zeitpunkt des Rename nur SHR, belegt lediglich 
das Member exklusiv. 

/* REXX 

PGM-Name..: #LM3 

Zweck.....: Rename eines Members der aktiven Datei 

*/ 

address ispexec 

"isredit macro (alt,neu)" 

if rc > 0 then do 

   zedsmsg="Parameter fehlt" 

   zedlmsg="Alten und neuen Membernamen angeben" 

   "setmsg msg(isrz000)" 

   exit 

end 

"lminit dataid(did) dataset(fb132.data) enq(shrw)" 

"lmopen dataid(&did) option(output)" 

"lmmren dataid(&did) member(&alt) newname(&neu)" 

if rc = 0 then do 

   zedsmsg="RENAME o.k." 

   zedlmsg=alt "wurde in" neu "umbenannt" 

end 

else do 

   zedsmsg="Fehler" 

   zedlmsg="Rename von" alt "nach" neu "nicht möglich" 

end 

"setmsg msg(isrz000)" 

"lmclose dataid(&did)" 

"lmfree dataid(&did)" 

exit 

3.1.7. Memberstatistiken 

Die statistischen Informationen für Members, welche vom Editor gepflegt werden, können im Be-
darfsfall auch per Programm ermittelt werden. Das folgende Beispiel zeigt auszugsweise, wie diese 
Informationen beschafft werden können. 

Achtung: Grundsätzlich können die Memberstatistiken natürlich auch durch ein Programm gesetzt 
werden – man darf längst nicht alles glauben was man sieht. 

/* REXX 

 

PGM-Name..: #LM4 

Zweck.....: Ist ein Member in einer Datei vor- 

            handen, werden die Memberstatistiken, 

            ansonsten wird ein Hinweis ausgegeben. 

*/ 

test = (arg(1) = "T") 

if test then trace "?r" 

address ispexec 

"lminit dataid(did) dataset(fb132.data) enq(shr)" 

"lmopen dataid(&did) option(input)" 

do forever 

   say "Membername ??" 

   pull mbr 

   if length(mbr) = 0 then leave 

   "lmmfind dataid(&did) member(&mbr) stats(yes)" 

   if rc = 0 then do /* Member verfügbar */ 

      Parse var zmldate 1 jj 3 . 4 mm 6 . 7 tt 9 . 

      mdate=tt"."mm"."jj 

      out=left(mbr,10) left(zlvers"/"zlmod,8) , 

         "Last Update:" mdate zlmtime "von" zluser 

   end 

   else do /*4(not available) 8(not found) */ 

      out="Member" mbr "nicht vorhanden" 

   end 

   say out 

end 

say "bye, bye..." 

"lmclose dataid(&did)" 

"lmfree dataid(&did)" 

Exit 

3.1.8. Ermitteln aller Member in einer Datei 

Die Namen aller Member werden als Wortkette aufgebaut und zurückgegeben. 

Wird das Programm als COMMAND aufgerufen, wird die Anzahl der Member ausgegeben. 

/* REXX * Memberliste mit LMMLIST */ 

Parse source . rxTYP rxNAME . 

arg dsn . 

address ispexec  

"control errors return" 

"lminit dataid(did) dataset('&dsn') enq(shr)" , 

   "org(po)" 

"lmopen dataid(&DID) option(input)" 

erg = 

"lmmlist dataid(&DID) option(list) member(MBR)" 

do while RC = 0 

   erg = erg strip(MBR) 

   "lmmlist dataid(&DID) option(list) member(MBR)" 

end 

"lmclose dataid(&DID)" 

"lmfree dataid(&DID)" 

If rxTYP <> "COMMAND" then return strip(erg) 

say rxNAME": Anzahl Member in" dsn ">>" words(erg) 

exit 

3.1.9. Setzen der Memberstatistiken 

Ermittelt alle Member einer Datei. Falls kein Update-User eingetragen ist, wird der Aufrufer einge-
tragen. 

/* REXX */ 

arg dsn  

address ispexec  

"control errors return" 

"LMINIT DATAID(DID) DATASET('&DSN') enq(SHR)" 

"LMOPEN DATAID(&DID) OPTION(INPUT)" 

"LMMLIST DATAID(&DID) OPTION(LIST)" , 

   "MEMBER(MBR) stats(YES)" 

do while RC = 0 

   if ZLUSER = "" then do 

      "LMMSTATS DATAID(&DID) MEMBER(&MBR)" , 

         "user("userid()")" 

   end 

   "LMMLIST DATAID(&DID) OPTION(LIST) MEMBER(MBR)" , 

      "stats(YES)" 

end 

"LMCLOSE  DATAID(&DID)" 

"LMfree  DATAID(&DID)" 

exit  

3.1.10. Liste von Dateinamen 

/* REXX  

   Ermittelt DS-Liste eines bestimmten Qualifiers 

   PARM___  WORTKETTE______________________________ 

   keine    dsn dsn ...  

   MITVOL   dsn vol dsn vol ... (ggfls Migrate-?) 

*/ 

 

parse source . RxTyp RxName . 

address ispexec 

"control errors return" 

if RxTyp="FUNCTION" then arg QUALIFIER,PARM,DUMMY 

else arg QUALIFIER PARM DUMMY 

/************************************************** 

* Gefundene DSNs als Wortkette in FOUND_DSN      

**************************************************/ 

FOUND_DSN = "" 

"lmdinit listid(LID) level("QUALIFIER")" 

"lmdlist listid(&LID) dataset(SDSN) stats(YES)" 

do while RC = 0 

   if ZDLVOL <> "*ALIAS" then do 

      FOUND_DSN = FOUND_DSN SDSN 

      if PARM = "MITVOL" then do 

         if ZDLVOL = "MIGRAT" then do 

            DMY = outtrap(LISTC.) 

            address tso , 

            "listc entry('"SDSN"') ALL" 

            DMY = outtrap("OFF") 

            parse value word(LISTC.11,2) with . , 

               "'" HEX_STR "'" 

            if HEX_STR = "78048083" then , 

               ZDLVOL = "Migrate-2" 

            else ZDLVOL = "Migrate-1" 

         end 

         FOUND_DSN = FOUND_DSN ZDLVOL 

      end 

   end 

   "lmdlist listid(&LID) dataset(SDSN) stats(YES)" 

end 

"lmdfree listid(&LID)" 

return strip(FOUND_DSN) 

4. ISPF Dialog Management Services 

Die Servicepalette des Dialog-Managers umfasst mehrere Bausteine, die im Verbund  das Erstellen 
von Dialog-Anwendungen sehr komfortabel gestalten. 

4.1. Die Services des Dialog Managers 

4.1.1. Display Service 

wird zur Anzeige von Bildschirmmasken genutzt. Dieser Service wird im Allgemeinen zur Datener-
fassung oder Präsentation genutzt. 

4.1.2. Selection Service 

Auch hier kann eine Bildschirmmaske angezeigt werden. Im Unterschied zum Display Service han-
delt es sich um so genannte Auswahlmenüs (vgl. Menü 0 aus ISPF/PDF), bei dem durch Eingabe 
Kürzels nachfolgende Aktivitäten aus dem Panel heraus gesteuert werden. Aktivitäten, die der Sel-
ection Service durchführt sind die Aufrufe von 

▪ Clisten oder REXX-Programme 
▪ Lademodulen 
▪ Auswahl-Bildschirmmasken 

4.1.3. Table Service 

Mit dem Tabellen Service wird ein Werkzeug angeboten, das Tabellenverarbeitung auf einer sehr 
hohen Komfortstufe ermöglicht. Bei kleineren Anwendungen (bis etwa 20.000 Datensätze) ersetzt 
dieser Service im Regelfall den Einsatz von Datenbanken. 

Die einzelnen Felder der Tabellensätze müssen weder in ihrer Länge, noch in ihrem Datentyp iden-
tisch sein. Zudem können im Nachhinein für einzelne oder alle Tabellensätze Felder angefügt wer-
den, die in ihrem Aussehen ebenfalls vollkommen flexibel sein können. 

Durch die Definition von Schlüsselfeldern können doppelte Datensätzen ausgeschlossen werden. 
Die Prüfung auf Vorhandensein übernimmt der Service selbst. 

Über Sortierbefehle kann die Tabelle nach beliebigen Feldern sortiert aufbereitet werden. 

Die wesentlichen Funktionen des Tabellen Service sind: 

▪ Anlegen und Löschen von Tabellen 
▪ Einfügen und Löschen von Tabellensätzen 
▪ Anzeige 
▪ Sequentielles Lesen 
▪ Suchvorgänge 

4.1.4. File Tailoring Service 

bietet im Wesentlichen die Möglichkeit, vorbereitete Datengerippe zusammenzuführen und in eine 
gemeinsame Datei auszugeben. Dabei können sowohl Variable substituiert, als auch Tabellen des 
Dialog Managers implementiert werden. 

Eine klassische Anwendung des File Tailoring wäre beispielsweise die Druckausgabe einer DMS-
Tabelle (gegebenenfalls auch nur selektierte Sätze) durch JCL, die durch das File Tailoring vervoll-
ständigt wird. 

Variablen Service 

Der Variablen Service bietet die Möglichkeit, die Inhalte beliebiger Variabler über Programmende 
oder über LOGOFF hinaus festzuhalten und zu einem beliebigen späteren Zeitpunkt wieder zur Ver-
fügung zu haben. Für das Ablegen von Variablen nutzt der Dialog Manager drei verschiedene Vari-
ablen-Pools: 

▪ Function Pool 
▪ Shared Pool 
▪ Profile Pool 

Darüber hinaus erlaubt er eine sehr komfortable Kommunikation zwischen Programmen (Variablen-
austausch), auch wenn diese nicht innerhalb der gleichen Task im TSO-Adressraum laufen. 

4.2. Dateilandschaft 

Die Dienste des Dialog Managers werden als Member von PO-Dateien erstellt.  

Die Aufteilung nach Zugehörigkeit zu einem Service und die Nutzung nachfolgender Dateinamen ist 
eine Empfehlung: 

4.2.1. userid.????????.PANELS 

Bildschirmmasken für die Services DISPLAY und SELECT 

4.2.2. userid.????????.TABLES 

Tabellenein- und -ausgaben für alle Tabellen-Services (wobei Ausgaben prinzipiell wahlfrei zuge-
ordnet werden können). 

4.2.3. userid.????????.MSGS 

Nachrichten, die in Bildschirmmasken benutzt werden. 

4.2.4. userid.????????.SKELS 

Skelette für den FILE-TAILORING-SERVICE. 

4.2.5. userid.????????.PROFILE 

Bildet mit seinen Membern den applikationsabhängigen Profile-Variablenpool ab. 

4.3. Begriffe 

Die einzelnen Elemente eines Dialogs unter Dialog Manager werden unter folgenden Begriffen zu-
sammengefasst, wobei nicht immer alle Elemente vorhanden sein müssen: 

4.3.1. Funktionen 

sind Programme, die im Dialog aufgerufen werden. Sehr gängig, weil sehr komfortabel zu schreiben, 
ist hier die Kodierung in 

▪ REXX oder 

▪ CLIST 

aber auch die klassischen Programmiersprachen 

▪ PL/1 

▪ COBOL 

▪ FORTRAN 

▪ PASCAL 

▪ APL2 und 

▪ ASSEMBLER 

sind möglich. Ein Dialog kann prinzipiell beliebig viele Funktionen enthalten, die in einer der oben 
angeführten Sprachen geschrieben sind. 

4.3.2. Paneldefinitionen 

sind die Formatbeschreibungen in Form kodierter Anweisungen über das Aussehen einer Bild-
schirmmaske. Ebenso liegen dort Anweisungen, die vor Anzeige einer Maske durchlaufen werden 
und die innerhalb der Maske durchzuführenden Prüfungen. 

4.3.3. Message Definitionen 

enthalten das Format und die Meldungen, die in Bildschirmmasken erscheinen sollen. 

4.3.4. Tabellen 

können im Dialog angelegt und manipuliert werden und werden in Dateien abgelegt. 

4.3.5. Datenskelette 

sind Fragmente beliebigen Inhalts (Daten, JCL-Sätze etc.), die zu einer Ausgabedatei zusammen-
gefügt werden. 

4.3.6. Variablen 

werden in der Regel als Bindeglied zwischen den einzelnen Elementen des Dialogs genutzt und 
steuern über ihre Inhalte im Allgemeinen den Verlauf des Dialoges. 

Die Dienste des Dialog-Managers sind nur unter ISPF verfügbar. Startet eine ISPF-Anwendung aus 
TSO-native, muss ISPF nachgestartet werden.  

Die Zuordnung kann über den TSO-Befehl ALLOCATE, oder, soweit es sich um erweiterte Zuord-
nungen über den Standard hinaus handelt, über die LIBDEF-Funktion des Dialog Managers erfol-
gen. 

4.4. DD-Namen und ihre Bedeutung 

DD-Name Bedeutung 

SYSPROC Programmbibliothek für CLISTen oder REXXe 

SYSEXEC REXX-Programmbibliothek 

ISPLLIB Lademodulbibliothek 

ISPPLIB Bibliothek für Panels (Bildschirmmasken). 

ISPMLIB Bibliothek für Nachrichten (Messages), die in Bildschirmmasken verarbeitet 
werden. 

ISPPROF Bibliothek mit den applikationsabhängigen PROFILE-Pools für Variable. 
Standardgemäß werden von ISPF/PDF folgende Member benutzt: 

• ISRPROF ISPF-Standard-Profile-Pool 

• ISREDIT Last-Qualifier-Dateiprofil (maximal 25) 

• ISREDRT Recovery-Tabelle für EDIT (maximal 5) 

• ISRCMDS ISPF-Kommandotabelle  

ISPTLIB/ISPTABL Eingabe/Ausgabe für Tabellenverarbeitung 

ISPSLIB/ISPFILE Eingabe/Ausgabe für Skelett-Dateien 

ISPILIB Graphik-Bibliothek für GIF-Dateien (Image-Lib) 

 

4.5. Dialoge 

Je nach Anforderung können unterschiedliche Arten von Dialogen aufgebaut werden. Grundsätzlich 
unterscheiden wir zwischen  

• einfachen Dialogen  
aus einem Programm heraus werden ein oder mehrere Bildschirmmasken aufgerufen 

• komplexen Dialogen  
hierarchisch geordnet werden mehrere Programme und Bildschirmmasken genutzt, die in 
der Regel über den SELECT-Service miteinander verbunden sind. 

Um mit den Routinen des DMS unter TSO arbeiten zu können, gibt es zwei Möglichkeiten, welche 
davon abhängen, ob ISPF bereits geladen ist, oder man sich im TSO-READY-Modus (TSO-native) 
befindet. 

• ISPF aktiv  
Alle Serviceroutinen sind verfügbar. Meist werden die Zuweisungen aller benötigten Biblio-
theken über den LIBDEF-Service veranlasst, der im hierarchisch obersten Programm liegen 
sollte. 

• TSO-READY 
Dialog Manager muss über das ISPSTART-Kommando geladen werden (siehe unter 
ISPSTART in der Syntaxbeschreibung  

Ob ISPF aktiv ist oder nicht, kann unter REXX erfragt werden: 

 IF SYSVAR(SYSISPF)="ACTIVE" THEN DO 

  [ISPF ist geladen] 

 END 

 ELSE DO 

  [ISPF ist nicht aktiv] 

 END 

oder 

 ADDRESS TSO "ISPQRY"  

 IF RC = 0 THEN DO 

  [ISPF ist geladen] 

 END 

 ELSE DO 

  [ISPF ist nicht aktiv] 

 END 

4.6. Return Codes und Fehlerbehandlung 

Endet ein ISPF-Service, reicht er einen Return Code an das aufrufende Programm zurück. Die mög-
lichen Return Codes sind 0, 4, 8, 10, 12, 16 und 20. Der jeweilige Hintergrund ist abhängig vom 
Service und kann bei der dazugehörenden Syntaxbeschreibung nachgelesen werden. 

In einem REXX-Programm liegt der Return Code in der Variablen RC, bei CLIST in &LASTCC vor, 
sobald die Steuerung  von DMS wieder an das Programm zurückgegeben wird. 

Tritt im Service ein RC < 12 auf, wird mit dem nächsten Befehl die Verarbeitung fortgesetzt. 

Bei einem Return Code größer oder gleich 12  ist die weitere Verarbeitung abhängig von der Ein-
stellung des Control-Statements im Dialog Manager: 

"CONTROL ERRORS CANCEL" 

Es erfolgt ein uneingeschränkter Abbruch des gesamten Dialoges bis hin zu ISPF oder TSO-
READY. 

"CONTROL ERRORS RETURN" 

Es wird in jedem Fall der nächste Befehl nach dem RC-verursachenden Service angesprungen. Im 
Allgemeinen ist dies die sicher sinnvollere Lösung, da in jedem Fall bei auftretenden Fehlern ein 
ungeregeltes Dialogende vermieden werden kann. 

4.7. Edit MODEL und MODEL CLASS 

Um bei der Entwicklung eines Dialoges möglichst viel Komfort zu haben, empfiehlt es sich, mit dem 
EDIT-Befehl MODEL zu arbeiten. Über die Befehlskombination "MODEL-A/B" werden die Befehle 
der jeweils aktiven Model-Klasse angezeigt. Nach Auswahl eines Befehles wird die Befehlssyntax 
mit diversen Erläuterungen (Notes) und den jeweiligen Return Codes an der vorgesehenen Stelle 
eingefügt. Wird im Dateiprofil "NOTES OFF" gesetzt, werden die Erläuterungen nicht angezeigt. Soll 
eine andere Model-Klasse angesteuert werden, als die derzeit aktuelle, kann diese über den Befehl 
"MODEL CLASS" eingestellt werden. 

4.8. Dialog Test 

Um einen Dialog zu testen, sollte grundsätzlich die PDF-Option 7 genutzt werden, da andernfalls 
beispielsweise Veränderungen an Panels nicht registriert werden (ein Panel wird beim ersten Aufruf 
in den Adressraum gelesen und nicht erneut von der Platte geladen, solange es im Speicher vor-
handen ist). 

Für das Austesten eines Dialoges bieten sich an: 

▪ 7.1 Select Services (Programme und Panels) 

▪ 7.2 Display Services (Panels und Messages) 

▪ 7.3 Manipulation des Profile-Variablenpools 

▪ 7.4 Arbeiten an DMS-Tabelle 

4.9. Variablendefinitionen 

Variablen im Dialog Manager können in unterschiedlichen Bereichen definiert und benutzt werden. 
Anwendung finden sie innerhalb von 

▪ Funktionen (zum Beispiel CLISTEN oder REXX) 

▪ Bildschirmmasken  

▪ Skeletons  

▪ Tabellenverarbeitung 

DMS unterscheidet dabei unter drei verschiedenen Bereichen, innerhalb derer Variable abgelegt 
sein können. Diese drei Bereiche werden zu unterschiedlichen Zeitpunkten gebildet und haben un-
terschiedliche Lebensdauer. Die Namen dieser Bereiche lauten 

▪ Shared-Pool und 

▪ Profile-Pool 

▪ Function-Pool 

Jede Funktion im Dialog (Programm, REXX oder CLIST) hat ihren eigenen Function-Pool. Er wird 
bei Start der Funktion initialisiert. Die Variablen sind nur innerhalb der Funktion verfügbar, die sie 
angelegt hat. Unter REXX oder CLIST haben auch externe UDF's keinen Zugriff auf diese Variablen. 

Wird innerhalb einer Funktion eine neue aufgerufen, wird der aktuelle Function-Pool abgedeckt und 
für die neue Funktion ein neuer Function-Pool zugewiesen. Wird eine Funktion beendet, wird der 
dazugehörende Function-Pool gelöscht. 

4.9.1. Shared-Pool 

Der Shared-Pool wird gebildet, wenn ISPF gestartet wird und besteht bis zum Ende von ISPF. Funk-
tionen können durch die Befehle VGET/VPUT auf den Shared-Pool zugreifen. Für den Zugriff auf 
den Shared-Pool ist zu beachten: 

Selection-Panels (Auswahl-Bildschirmmasken) greifen unmittelbar auf den Shared-Pool zu, da sie 
keinen Function-Pool kennen. 

Soll eine Variable aus dem Function-Pool auch im Shared-Pool abgebildet werden, ist sie mittels 
VPUT-Anweisung dorthin zu schreiben. 

Muss eine Variable aus dem Shared-Pool im aktuellen Function-Pool verfügbar sein, ist sie über 
VGET aus dem Shared-Pool zu lesen. 

4.9.2. Profile-Pool 

Variable, die über die Dauer einer ISPF-Session oder auch über LOGOFF hinaus festgehalten wer-
den sollen müssen im Profile-Pool abgebildet werden. Dieser Bereich ist in Form einer PO-Datei 
realisiert, innerhalb derer applikationsabhängige Variablen-Pools in Form von Membern eingetragen 
werden können. Die betreffende PO-Datei muss über das Symbol ISPPROF bekannt gegeben wor-
den sein. 

Die im Profile-Pool hinterlegten Variablen werden bei Beginn des Dialoges von Datei gelesen und 
im virtuellen Speicher abgebildet.  

Anders als der Shared- oder Function-Pool wird der Profile-Pool bei Dialogende vom virtuellen Spei-
cher auf Platte zurückgeschrieben. Über diesen Weg ist es beispielsweise auch dem ISPF-Editor 
möglich, sich zu merken, wie der aktuelle Wert für das Blättern am Bildschirm (SCROLL-AMOUNT) 
gesetzt war. 

Die Eintragungen in den Profile-Pool oder das Lesen daraus werden über die DMS-Befehle "VPUT" 
und "VGET" durchgeführt. 

Vor Änderung des Profile-Pools im Editor wird dringend abgeraten, da die Werte teils im Binärformat 
hinterlegt sind. Explizite Veränderungen im Profile-Pool sind über Menü 7 des ISPF durchzuführen. 

4.9.3. VPUT-Service 

Soll eine Variable in allen Funktionen des Dialogs verfügbar sein, muss sie durch VPUT in den 
Shared- oder Profile-Pool übertragen werden. Existiert dort bereits eine gleichnamige Variable, wird 
ihr Inhalt überschrieben. Um den Service gezielt zum Schreiben in einen bestimmten Variablenpool 
zu veranlassen, können entsprechende Schlüsselworte mitgegeben werden.  

4.9.3.1. ASIS  

Die Variable wird in demjenigen Pool (Shared oder Profile) hinterlegt, in dem bereits eine Variable 
gleichen Namens existiert. Wird sie in keinem der beiden Pools gefunden, wird sie in den Shared-
Pool eingetragen. 

4.9.3.2. SHARED  

Die Variable wird in den Shared-Pool eingetragen. 

4.9.3.3. PROFILE  

Die Eintragung erfolgt im Profile-Pool. Existiert eine gleichnamige Variable im Shared-Pool, wird sie 
dort ausgetragen. 

4.9.4. VGET-Service 

Durch VGET kann eine Variable, die nur im Shared-Pool oder Profile-Pool existiert gelesen und in 
den Function-Pool kopiert werden. 

Ist eine gleichnamige Variable im Function-Pool bereits vorhanden, wird ihr Inhalt überschrieben. 

4.9.5. Gewichtung der POOLS 

Generell wird von den ISPF-Services die Suchreihenfolge 

▪ Function-Pool 
▪ Shared-Pool 
▪ Profile-Pool 

eingehalten. Existiert eine Variable in mehr als einem Variablenpool (prinzipiell wäre dies auch mit 
unterschiedlichen Inhalten möglich), wird sie mit ihrem Inhalt aus dem Pool genommen, in dem sie 
zuerst gefunden wurde. 

Die Ausnahme bildet hier der SELECT-Service, der keinen Function-Pool kennt. Von ihm werden 
zwangsläufig nur Shared-Pool und Profile-Pool durchsucht 

4.10. Display Panels (Datenpanels) 

Unter einem Panel versteht man das Aussehen eines Bildschirmes (Layout), wie es sich für den 
Terminalbenutzer darstellt. Die Kodierung eines Panels enthält die Angaben zur Text- und Variab-
lenanordnug auf einer Bildschirmseite, sowie Definitionen über Initialisierungsarbeiten vor Anzeige 
und Plausibilitätsprüfungen vor Verlassen der Maske. 

Der Name der Maske ist gleichbedeutend mit dem Membernamen, unter dem sie in einer Datei unter 
ISPPLIB abgespeichert wird. 

4.10.1. Sections 

Das Panel gliedert sich in verschiedene Bereiche (Sections), die bei der Verarbeitung eine unter-
schiedliche Rolle spielen. Um welchen Abschnitt im Panel es sich handelt, wird mit dem so genann-
ten Sectionnamen bestimmt, der direkt hinter einer schließenden Klammer ab Spalte 1 kodiert wer-
den muss. 

Nicht alle Abschnitte sind zwingend notwendig. Die Reihenfolge ist zwingend. 

Section MUSS Beschreibung 

)INEXIT Nein Input Exit Section. Benennt ein Programm, welches die Steue-
rung bei Panelstart bekommt 

)CCSID Nein Code Character Set Identifier. Für Codepage-Translation (eher 
selten) 

)PANEL Nein Zeigt auf eine Keylist (PF-Tasten-Definition) und bestimmt, ob 
das Panel im CUA-Modus aufgerufen wird 

)ATTR Nein Definiert Feldattribute für )BODY, )AREA und )MODEL 

)ABC Nein ActionBarChoice. Die Aktionsleiste, sofern das Panel eine haben 
soll 

)ABCINIT Nur wenn )ABC 
vorhanden ist 

Initialisierungsbereich für eine Action Bar 

)ABCPROC Nein Verarbeitungsbereich für eine Action Bar 

)BODY Ja Enthält das, was der Anwender später sehen wird 

)MODEL Nur bei  
Tabellenpanels 

Definiert die Darstellung EINES Tabellensatzes und wird so oft 
wiederholt, als es der physische Screen zulässt 

)AREA Nein Definiert einen blätterbaren Bereich im Panel. Immer sinnvoll, 
wenn der Body mehr als 24 Zeilen hat, da es sonst bei Screen-
size 24/80 zum Abbruch beim Panelaufruf kommt 

)INIT Nein Initialisierungsteil. Wird vor Aufruf des )BODY durchlaufen 

)REINIT Nein Wird aufgerufen, wenn im )PROC-Teil Fehler erkannt werden 

)PROC Nein Verarbeitungssection. Feldprüfungen und dergleichen 

)FIELD Nein Beschreibt Felder, innerhalb derer links/rechts Scrolling möglich 
ist. 

)HELP Nein Paneldefinition für Feldhilfen 

)LIST Nein Beschreibt eine Liste, die im Panel gebildet werden soll 

)PNTS Nein Point and Shoot Section. Hier werden alle Point and Shoot Fel-
der definiert 

)END Ja Definiert das Panelende. Alle Zeilen, die danach kommen, wer-
den ignoriert 

 

Die Verarbeitung der einzelnen Sektionen im Panel geschieht nicht in einer linearen Reihenfolge 
von oben nach unten und wird in folgender Grafik verdeutlicht (Standardpanel): 

  



Inside z/OS Seite 221 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

4.10.2. Ablaufsteuerung Panel 

4.10.2.1. Vorbereitende Arbeiten 

Zunächst werden die Attributmerkmale interpretiert, sind keine Attribute definiert, erfolgt die Umset-
zung dem Standard entsprechend. 

Variablen werden entsprechend den Anweisungen des INIT-Teiles gefüllt. Darüber hinaus sind alle 
Variablen des Funktion-, Shared- und Profile-Pools verfügbar. 

Das Panel ist zu diesem Zeitpunkt noch nicht am Bildschirm sichtbar. 

4.10.2.2. Panel-Anzeige 

Das Panel wird am Terminal abgebildet. Nach Aufruf des Panels tritt die Funktion in einen Wartezu-
stand der entweder durch < ENTER> oder eine PF-Taste beendet wird. 

Ausnahme: 

Wird in der Kommandozeile des Panels (soweit überhaupt kodiert) ein gültiges ISPF-Kommando 
eingegeben, wird dies vom Dialog Manager ausgeführt und die steuernde Funktion im Wartezustand 
belassen. 

4.10.2.3. Panel-Auswertung 

Sinnvollerweise (wenn auch nicht zwingend) wird im Panel eine )PROC-Section kodiert. Dieser Be-
reich wird für die Auswertung der Dateninhalte angesprungen. Hier erfolgt im Regelfall die Plausibi-
litätsprüfung für alle Eingabefelder des Panels. 

Grundsätzlich können diese Prüfungen auch in der aufrufenden Funktion kodiert werden. Aus Per-
formancegründen ist aber eine Überprüfung im Panel soweit als möglich empfehlenswert. 

4.10.2.4. Erneute Panelanzeige 

Werden in der Plausibilitätsprüfung Fehler erkannt, wird in den )REINIT-Teil des Panels verzweigt. 
Hier können fehlerabhängige Modifikationen an der Bildschirmmaske vorgenommen werden. Im An-
schluss daran, oder wenn kein )REINIT-Teil kodiert wurde, wird das Panel erneut am Terminal ab-
gebildet (die Steuerung geht wieder zum )BODY-Teil). 

4.10.2.5. Ende der Panelanzeige 

Bei Erreichen der )END-Marke wird die Steuerung an die aufrufende Funktion zurückgegeben. Diese 
Rückgabe der Steuerung über den weiteren Ablauf erfolgt bei END, RETURN oder wenn im )PROC-
Teil keine Fehler erkannt wurden. 

4.10.3. )PANEL Section 

Die )PANEL Section wird benötigt, wenn ein Panel im CUA Modus laufen soll. Dies ist eine Erweite-
rung aus den 1990er Jahren, über welche man dem Bediener anheimstellen will, wie bestimmte 
Dinge im Panel angezeigt werden sollen. Nicht der Programmierer bestimmt beispielsweise die Far-
ben, sondern der Endanwender kann festlegen, wie bestimmte Feldtypen angezeigt werden sollen. 
Die Einstellungen kommen in ISPF/PDF unter <SETTINGS><COLORS><CUA ATTRIBUTES> vor-
genommen werden. Welche zusätzlichen Attributsangaben möglich sind, ist im "Band-2 Syntax-Re-
ferenz" beschrieben. 

Darüber hinausgibt es hier die Möglichkeit Bezug zu PF-Tastenbelegungen herzustellen. Das heißt: 
Ein Panel kann seine eigene PF-Tastenbelegung mitbringen. Es gibt allerdings nichts Schlimmeres, 
als, wenn man gewöhnt ist, über die PF12-Taste das RETRIEVE Kommando abzusetzen, Über die 

Panelspezifische Belegung hinter der PF-12-Taste den Befehl EXIT vorzufinden. Ich selbst definiere 
deshalb in meinen ISPF Settings, dass panelspezifische PF-Tasten ignoriert werden. 

4.10.4. )CCSID Section 

Kann einen Verweis auf Codepages liefern.  

Folgende CCSID-Angaben sind möglich: 

CCSID Character-
Set 

Code-
page 

Extended CCSID1 Unterstützung für Länder 

00037 679 37 USA, Kanada, Niederlande, Portugal, Brasilien, Australien, 
Neuseeland 

00273 697 273 Deutschland, Österreich 

00277 697 277 Dänemark, Norwegen 

00278 697 278 Finnland, Schweden 

00280 697 280 Italien 

00284 697 284 Spanien, Latain Amerika (spanisch) 

00285 697 285 Vereinigtes Königreich 

00297 697 297 Frankreich 

00420 235 420 Arabisch 

00424 941 424 Hebräisch 

00500 697 500 Schweiz, Belgien 

00838 1176 838 Thailand 

00870 959 870 Latin-2 

00871 697 872 Island 

00875 923 875 Griechenland 

00880 960 880 Kyrillisch 

01025 1150 1025 Kyrillisch 

01026 1026 1026 Türkei 

01047 697 1047 Latin-1 

01123 1326 1123 Ukraine 

 

CCSID Character-
Set 

Code-
page 

Extended CCSID2 Unterstützung (EURO) für Länder 

00924 1353 0924 Latin9 

01140 659 1140 USA, Kanada, Niederlande, Portugal, Brasilien, Australien, 
Neuseeland 

01141 659 1141 Österreich, Deutschland 

01142 695 1142 Dänemark, Norwegen 

01143 695 1143 Finnland, Schweden 

01144 695 1144 Italien 

01145 659 1145 Spanien, Latain Amerika (spanisch) 

01146 659 1146 Vereinigtes Königreich 

01147 659 1147 Frankreich 

01148 659 1148 Schweiz, Belgien 

01149 659 1149 Island 

01153 1375 1153 Latin2 

01154 1381 1154 Kyrillisch 

01155 1378 1155 Türkei 

01158 1388 1158 Ukraine 

01160 1395 1160 Thailand 

04899 1356 0803 Hebräisch 

04971 1371 0875 Griechenland 

12712 1357 0424 Hebräisch 

16804 1461 0420 Arabisch 

 

Auf die Angaben für traditionell Chinesisch und dergleichen möchte ich verzichten 

4.10.5. )ATTR Section 

Im Attribute-Bereich können zunächst die unterschiedlichen Attribute der im Panel angezeigten Fel-
der (Text, Output oder Input) abweichend zum Standard geregelt werden. Fehlt der )ATTR-Teil, 
werden die Standardzeichen benutzt, die folgende Bedeutung haben: 

 % Text intensiv 
 + Text normal 
 _ Inputfeld intensiv 

Diese Attribute müssen im )BODY-Teil dem Feld vorangestellt werden, für das sie gelten sollen. 
Diese Definition für ein Feld gilt so lange, bis das nächste Attributzeichen erkannt wird. 

Soll eines oder mehrere der Standardattribute als Text genutzt werden, sind die Default-Werte neu 
zu setzen. Dies geschieht im Regelfall in der Kopfzeile des )ATTR-Teiles, oder falls dieser fehlt in 
der Kopfzeile des )BODY-Teils in folgender Form: 

 )ATTR DEFAULT(x1 x2 x3) oder 

 )BODY DEFAULT(x1 x2 x3) 

Falls DEFAULT kodiert wird, müssen immer alle drei Standardzeichen angegeben werden, wobei 
die Reihenfolge strikt einzuhalten ist und die Bedeutung immer der Reihe nach 

1. Text intensiv 
2. Text normal und 
3. Eingabefeld intensiv 

erfolgt. 

Werden im Panel weitere Feldattribute abweichend zu den standardgemäßen benutzt, müssen 
diese im )ATTR-Teil definiert werden. Das Attributzeichen kann zweistellig-HEX oder als Character-
zeichen dargestellt werden. Die Zeichen 

&, blank '40'x, NUL '00'x, ShiftOut '0E'x, ShiftIn'0F'x 

dürfen nicht benutzt werden. Die Eigenschaften der Felder werden durch Schlüsselworte bestimmt. 
Die Wertstellungen der einzelnen Schlüsselworte kann durch Literal oder Variable festgesetzt wer-
den (Ausnahme: TYPE(TEXT) muss kodiert werden wie geschrieben). 

4.10.5.1. Attribut-Defaults 

Die Standardwerte bei einem Panelaufruf sind bei 

 TYPE(INPUT) CAPS(ON) JUST(LEFT) PAD(USER) 

 TYPE(OUTPUT) CAPS(ON) JUST(LEFT) PAD(' ') 

4.10.5.2. Attribut-Schlüsselworte 

Die gängigen Attribut-Schlüsselworte sind: 

Keyword Bedeutung 

AREA Bereichsdefinitionen innerhalb der Maske 

TYPE bestimmt die Art des Feldes 

INTENS Steuert die Helligkeit 

CAPS Groß-/Kleinschreibung 

JUST Ausrichtung der Daten 

PAD Füllzeichen für Ein-/Ausgabefelder 

PADC Abhängiges Füllzeichen (Conditional) 

SKIP Textfelder werden übersprungen 

COLOR Weist dem Feld eine Farbe zu 

Aus nachfolgendem Beispiel kann der Aufbau einer Bildschirmmaske mit diversen Feldattributen 
ersehen werden: 

)ATTR DEFAULT(@+_) 

 * TYPE(OUTPUT) INTENS(HIGH) CAPS(OFF)  

 ^ TYPE(INPUT) INTENS(LOW) COLOR(BLUE) 

 $ TYPE(TEXT) INTENS(LOW) 

 # TYPE(TEXT) INTENS(HIGH) COLOR(YELLOW) 

)BODY 

#======== Kunden-Abfrage ============================= 

@ 

$Datum:*ZDATE                            $Zeit: *ZTIME 

@ 

@ 

$ Eingabe: Kundennummer 

$ Ausgabe: Adresse und Rabattsatz 

@ 

@ 

# Kundennummer  ==> ^KDNR  # 

# Name    ==> *NAME 

# Strasse   ==> *STR 

# PLZ/ORT   ==> *PLZ   *ORT 

# Rabatt   ==> *RAB@% 

# 

# 

$ ENDE mit #END 

$ Speichern #ENTER 

)END 

4.10.5.2.1. TYPE 

Das Schlüsselwort TYPE beschreibt die unterschiedlichen Formen der Darstellung der einzelnen 
Felder in der Maske.  

4.10.5.2.2. TYPE(TEXT) 

zeigt ein Feld exakt so, wie es im )BODY-Teil geschrieben ist. Werden Variablen (führendes &) im 
Textteil genutzt, erscheint am Bildschirm an der entsprechenden Stelle der Variableninhalt. Werden 
Variablen benutzt, ist die Ausgabezeilenlänge flexibel. 

4.10.5.2.3. TYPE(INPUT) 

Eingabefeld. Diese Stelle in der Maske ist beschreibbar. Der eingegebene Wert wird bei <ENTER> 
in eine Variable übertragen und im Variablenpool abgestellt. Die Länge der Felder ist abhängig vom 
nächsten nachfolgenden Attributzeichen. 

4.10.5.2.4. TYPE(OUTPUT) 

Ausgabefeld. Der Inhalt wird einer Variablen entnommen und angezeigt. Diese Felder sind nicht 
beschreibbar. Im Unterschied zu Variablen in TEXT-Feldern wird hier unter Umständen nicht der 
gesamte Inhalt einer Variablen, oder aber mehr (abhängig von der Längendefinition des Output-
Feldes) angezeigt. 

4.10.6. )BODY Section 

Im )BODY-Teil werden die Feldnamen mit Attributzeichen, aber ohne Ampersand geschrieben (Aus-
nahme ist eine Variable in einem Textfeld). 

4.10.6.1. Gestaltung der Panels 

Im )BODY-Teil wird die Bildschirmmaske so abgebildet, wie sie für den Benutzer am Bildschirm 
erscheinen wird. Dieser Teil darf deshalb in Punkto Spalten- und Zeilenzahl die Werte eines Termi-
nals nicht überschreiten. 

In Übereinstimmung zu den PDF-Panels  sollte der Aufbau innerhalb gewisser Normen stattfinden: 

▪ Durch den Befehl PANELID ON erscheint in Zeile 1, Stelle 1 bis 8 der Panelname 
▪ In den maximal letzten 24 Byte der ersten Zeile erscheint im Bedarfsfall eine kurze Nachricht 

(Short Message). 
▪ Der verbleibende Rest der ersten Zeile sollte für eine Überschrift genutzt werden. 
▪ In der zweiten Zeile steht links das COMMAND-Feld und rechts das SCROLL-Feld (für Blät-

terfunktionen bei Tabellenanzeige). 
▪ Die dritte Zeile sollte generell freigelassen werden. Hier erscheint eine Nachricht (Long Mes-

sage), wenn auf eine kurze Nachricht hin die HELP-Funktion aufgerufen wird. 

Der Rest des Bildschirmes steht zur freien Verfügung. 

4.10.6.2. Pfeilmarkierungen (CUA: Leader Dots) 

In den Panels des PDF werden Eingabefeldern in aller Regel  Pfeile vorangestellt. Diese Vorgehens-
weise wird auch bei eigenen Dialogen aus zweierlei Gründen empfohlen: 

▪ Eingabefelder werden visuell sofort erkannt, da sie das gewohnte Erscheinungsbild aus PDF 
aufweisen. 

▪ Der Pfeil als Feldmarkierung erlaubt auch innerhalb des eigenen Dialoges Menüsprünge 
über das letzte vorgelagerte Primary-Menü (vgl.: Direktsprung von EDIT nach BROWSE 
durch =1). 

Ausnahmen 

Manche Bildschirmmasken zwingen aufgrund bestimmter Vorgaben zu einer Abweichung vom Stan-
dard-Layout. In diesem Fall sieht die Kodierung etwas anders aus, wie die folgende Seite zeigt. 

)BODY DEFAULT(#+_)  CMD(variable) 

    SMSG(variable) LMSG(variable) ASIS 

    EXPAND(//)     WIDTH(integer) 

    WINDOW(spalten,zeilen) 

+/-/< Überschrift >/-/ 

4.10.6.3. Parameter in der )BODY Section 

▪ DEFAULT gleichbedeutend wie in )ATTR 

▪ CMD, SMSG und LMSG müssen mit Variablennamen versorgt werden, wenn die Standorte 
der Felder abweichend vom Standard an beliebiger Stelle der Bildschirmmaske liegen sollen. 

▪ ASIS veranlasst, dass die Standorte des COMMAND-Feldes und der Nachrichtenfelder nicht 
durch die PDF-Option 0.4 beeinflusst werden. 

▪ WIDTH regelt die Spaltenbreite grafikfähiger Terminals, die die Breite von 80 überschreiten 
können. Die Angabe ist zwingend, wenn das Terminal über die Spalte 80 hinaus genutzt 
werden soll. 

▪ EXPAND erlaubt das Festlegen so genannter Begrenzer, zwischen denen ein Füllzeichen 
gesetzt werden kann. Mit diesem Füllzeichen lassen sich Texte links- oder rechtsseitig bis 
Zeilenende auffüllen. 

▪ WINDOW definiert die Fenstergröße, in dem das Panel als PopUp dargestellt wird. Die An-
zeige des Fensterpanels variiert je nach Aufruf. 

4.10.6.3.1. Panelaufruf als Window mit REXX 

Die WINDOW-Angabe wird ignoriert, wenn der DISPLAY-Service ohne vorhergehenden ADDPOP 
aufgerufen wird. Ohne WINDOW() im Panel, werden, sobald ADDPOP einmal genutzt wurde, die 
folgenden Panels im zuletzt definierten Fenster gezeigt. Wird WINDOW() nicht kodiert und der ADD-
POP-Befehl genutzt, wird das Panel ganzseitig mit Rahmen abgebildet. Die Parameter POPLOC(), 
ROW() und COLUMN() werden ignoriert. 

4.10.6.3.2. Panelaufruf als Window aus DMS 

Ist einmal ein Fenster definiert, erscheinen alle nachfolgenden Panels als Fensterpanel in der Größe 
der zuletzt aufgerufenen Bildschirmmaske, unabhängig davon, ob das Panel eine WINDOW()-An-
gabe im )BODY-Teil enthält, oder nicht. 

Mit der Systemvariablen ZWINTTL kann ein Text angegeben werden, der in der oberen Leiste des 
Fensterrahmens als Titel zentriert angezeigt wird. 

4.10.7. Scrollable Panels 

Ein häufiges Problem ist die physische Größe des Bildschirmes. Der )BODY-Teil darf grundsätzlich 
nur so viele Zeilen lang sein, als auf dem Terminal Platz haben. Oftmals können aber nicht alle 
benötigten Felder auf einer Seite abgebildet werden. Generell könnte dieses Problem auf der Funk-
tionsseite über die Bearbeitung mehrerer Panels gelöst werden. Diese Verfahrensweise macht eine 
Dialoganwendung aber kompliziert und änderungsunfreundlich (auch wenn dies in der Praxis ange-
troffen wird). 

Ein anderes Problem sind oftmals auch die lieben Kollegen. Obwohl die Bildschirme immer größer 
werden, gibt es immer noch viel zu viele Mitarbeiter, welche auf einer 43" Monitor ihre 3270-Emula-
tion in 24/80 fullscreen betreiben. Und NEIN, es liegt meist nicht daran, dass sie nicht wissen, wie 
man das ändert, sondern, dass sie zum Arbeiten dann keine Brille brauchen. Wirklich. Gibt es. 

Abhilfe schafft ein Bereich innerhalb des Panels, in dem über UP und DOWN geblättert werden 
kann. Der Bereich kann von seiner Größe her im Panel fest definiert werden, oder sich situations-
abhängig selbstständig bis zum unteren Panelrand ausdehnen. Innerhalb des Bereiches können 
beliebige Informationen oder Felder liegen. 

Prinzipiell können mehrere derartige Bereiche in einem Panel definiert werden. Mit jedem der Berei-
che kann im Grunde ein Panel im Panel dargestellt werden, das angesteuert werden kann, ohne die 
eigentliche Bildschirmmaske verlassen zu müssen. 

Ist nur ein solcher Bereich vorhanden, kann innerhalb dessen mit UP oder DOWN geblättert werden, 
unabhängig davon, wo der Cursor aktuell steht. Bei mehr als einem Bereich erfolgt das  Blättern in 
dem Bereich, in dem der Cursor positioniert ist. 

4.10.7.1. )AREA bereichsname 

Ist eine Section, die nur benötigt wird, wenn mit einer SCROLL-AREA gearbeitet werden soll. be-
reichsname ist der Name des Bereiches, der in der )AREA-Section beschrieben werden soll. 

Für jede SCROLL-AREA im )BODY-Teil des Panels muss eine )AREA-Section kodiert werden, die 
den gleichen Namen trägt, wie er im )BODY-Teil angegeben wurde. 

Im nachfolgenden Beispiel wird ein Panel gezeigt, das Hintergrundinformationen (weniger wichtige) 
in einer SCROLL-AREA führt.  

)ATTR DEFAULT(@+_) 

 § TYPE(INPUT) JUST(RIGHT) PAD(0) 

 ^ TYPE(INPUT) JUST(RIGHT) PADC('_') CAPS(OFF) 

 # AREA(SCRL) EXTEND(OFF) 

)BODY EXPAND($$) 

@--------< Kunden-Erfassung >$-$ 

@COMMAND ==>_ZCMD 

@ 

@ 

+  Kundennummer ==>§KDNR + 

+  Name         ==>^NAME          + 

+  Vorname      ==>^VNAME         + 

+  Strasse      ==>^STR           + 

+  PLZ/Ort      ==>^PLZ  ^Ort                 + 

+ 

@  Weitere Angaben: 

+ 

#AREA1                                             # 

#                                                  # 

#                                                  # 

#                                                  # 

#                                                  # 

#                                                  # 

+ 

@  Ab: PF8   Auf: PF7 

+ 

)AREA AREA1 

+Rabatt        ==>§RAB+% 

+Telefonnummer ==>^TELNR      + 

+Faxnummer     ==>^FAXNR      + 

+Anrede        ==>^ANRED+  (Herr, Frau, oder Firma) 

+Sprache       ==>^S+      (D - F - I - E) 

+Kunde seit    ==>^SEITDAT   + 

+Hausbank      ==>^BANKNAME     + 

+Kontonummer   ==>^KONTONR      + 

+Vertreter     ==>^VERTRET    + 

+Interne Nr.   ==>^INTERN + 

+Interner Code ==>^CODE +(numerisch) 

+Werbung       ==>^WERB+ (Ja oder Nein) 

+Bemerkung 1   ==>^COMM1               + 

+Bemerkung 2   ==>^COMM2               + 

)PROC 

   VER(&NAME,NB,MSG=FAKT001) 

   VER(&VNAME,NB,MSG=FAKT002) 

   VER(&RAB,NB,NUM,MSG=FAKT012) 

   VER(&ANRED,LIST,'Herr','Frau','Firma',MSG=FAKT021) 

   VER(&S,LIST,'D','E','F','I',MSG=FAKT022) 

   VER(&CODE,NB,NUM,MSG=FAKT023) 

   VER(&WERB,LIST,'JA','NEIN',MSG=FAKT024) 

)END 

4.10.8. )HELP Section (Feldhilfen) 

Hier wird ein HELP-Panel definiert, das feldbezogen bei Drücken der PF1-Taste erscheint. Je Feld 
im BODY-Teil kann ein HELP-Panel definiert werden. Im Beispiel hat das INPUT-Feld für Rabatt ein 
HELP-Panel: 

: 

# PLZ/ORT  ==> *PLZ   *ORT 

# Rabatt   ==> *RAB@% 

)HELP 

   FIELD(RAB) PANEL(hpname) 

)END 

4.10.9. )INIT, )REINIT, )PROC (Panel-Aktivitäten) 

In allen drei Bereichen können bestimmte Informationen aus einem Panel verarbeitet werden. Da 
jeweils die gleichen Sprachelemente in diesen Sektionen erlaubt sind wird bei der Beschreibung der 
Möglichkeiten im Folgenden nicht weiter unterschieden. 

Grundsätzlich finden Befehle, Funktionen und Steuervariable Anwendung, wobei nicht in jedem Be-
reich die Verwendung aller Elemente sinnvoll ist. 

4.10.9.1. Befehle 

werden für die Steuerung des Ablaufes genutzt. Folgende sechs Befehle können kodiert werden: 

Befehl Bedeutung 

VGET Lesen Variable aus Variablenpool 

VPUT Schreiben Variable in Variablenpool 

IF / ELSE Entscheidungsfindung 

VER Feldprüfung (Verify) 

REFRESH Feld-Aktualisierung 

&var = wert Variablenzuweisung 

*REXX REXX-Logik im Panel 

4.10.9.2. Funktionen 

sind kleine Befehlsprozeduren oder Programme, die den Bedürfnissen des Benutzers entsprechend 
eine Wertrückgabe bewirken. Funktionen sind beispielsweise: 

Funktion Bedeutung 

TRANS Zeichenfolgen übersetzen 

TRUNC Zeichenfolgen abschneiden 

PFK Definition der Funktionstasten 

4.10.9.3. Steuervariablen 

sind Variablen, die (wie ihr Name sagt) steuernde Funktionen in einem Panel ausführen. Beispiels-
weise kann die Cursor-Positionierung in einer Bildschirmmaske individuell geregelt werden. Einige 
der Steuervariablen sind: 

Steuervariable Bedeutung 

.ALARM Ausgabe Warnton 

.CURSOR Cursorpositionierung auf Feld 

.CSRPOS Cursorpositionierung innerhalb eines Feldes 

.MSG Verweis auf eine definierte Nachricht 

.TRAIL Restzeichenfolge nach TRUNCate 

 

Einige der gängigsten Elemente soll das nachfolgende Beispiel demonstrieren: 

)ATTR DEFAULT(@+_) 

   $ TYPE(OUTPUT) INTENS(HIGH) CAPS(OFF)  

   ^ TYPE(INPUT) INTENS(LOW) PADC('.') 

)BODY EXPAND(\\) 

@\*\ Kunden-Erfassung \*\ 

@COMMAND ==>_ZCMD                                       + 

@ 

+Kundennummer ==>$KDNR  + 

+Name         ==>^NAME                    + 

+Vorname      ==>^VNAM                    + 

+Strasse      ==>^STR                      + 

+PLZ/ORT      ==>^PLZ  ^ORT                          + 

+Rabatt       ==>^RAB@% 

)INIT 

   &NAME=&Z 

   &VNAM=&Z 

   &STR=&Z 

   &PLZ=&Z 

   &ORT=&Z 

   &RAB=0 

   .CURSOR=NAME 

)REINIT 

   .ATTR(.CURSOR)='INTENS(HIGH) COLOR(RED)' 

)PROC 

   VER(&NAME,NB,MSG=MSG001) 

   VER(&VNAM,NB,MSG=MSG001)  

   VER(&VNAM,LEN,'<=',2,MSG=MSG001) 

   VER(&STR,NB,MSG=MSG001) 

   VER(&PLZ,NB,NUM,MSG=MSG002) 

   VER(&PLZ,RANGE,1000,8999,MSG=MSG003) 

   VER(&ORT,NB,MSG=MSG001) 

   VER(&RAB,LIST,0,5,10,20,MSG=MSG004) 

)END 

4.10.9.4. Wertzuweisungen 

Mit den Wertzuordnungen wird einer Variablen ein Wert zugewiesen. Dazu kann eine Zeichenfolge 
oder eine Variable dienen. 



Inside z/OS Seite 232 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

Wird eine Zeichenfolge zugewiesen sollte diese zwischen Apostrophe eingegrenzt sein. Zwingend 
ist dies, wenn die Zeichenfolge Sonderzeichen (auch Umlaute oder ß) oder Blanks enthalten soll. 

Wird für die Zuweisung eine Variable genutzt, wird deren Inhalt übertragen. Bei der Auflösung von 
Variablen kommt es allerdings nur zu einer einmaligen Substitution (zeigt der Inhalt einer Variablen 
auf den Namen einer anderen Variablen, kann deren Inhalt nicht ermittelt werden). 

Anders als bei Wertzuordnungen in REXX kann im Panel bei der Wertzuweisung keine Arithmetik 
aufgelöst werden. Erforderliche Berechnungen müssen in die aufrufende Funktion verlagert werden. 

4.10.9.5. Feldprüfungen 

Durch VERIFY kann der Inhalt einer Variablen überprüft werden. Die Prüfungen erfolgen in der Rei-
henfolge der VER-Instruktionen. Endet die Prüfung negativ, wird ein Standardtext, oder eine vorbe-
reitete Meldung ausgegeben. Anschließend wird (sofern vorhanden) der )REINIT-Teil angesprun-
gen, bevor wieder in den )BODY verzweigt wird. Der CURSOR wird in das reklamierte Feld gestellt. 

Bei positivem Ergebnis ist sichergestellt: 

VERIFY Variableninhalt 

NB Ungleich BLANK 

RANGE,7,25 Numerisch zwischen 7 und 25 

LIST,A,B,Z Einer der drei angegebenen Werte 

LISTV Die zu prüfenden Werte liegen in einer Variablen 

LEN,GT,7 Der Inhalt (Feldlänge) ist mindestens 8 Byte 

4.10.9.6. Re-Initialisierung 

Sofern der )REINIT-Teil kodiert ist, wird er durchlaufen, wenn im )PROC-Teil Fehler erkannt wurden, 
um vor erneuter Präsentation der Bildschirmmaske Veränderungen in Variablen oder Attributen zu 
erreichen. 

Im zugrunde liegenden Beispiel wird bei Erkennen eines Fehlers das betreffende Feldattribut geän-
dert, so dass das fehlerhaft ausgefüllte Feld auch optisch unter den restlichen Feldern hervortritt. 

Um die Möglichkeiten der Ablaufsteuerung und Feldauffrischung zu demonstrieren wird unser Bei-
spiel entsprechend erweitert: 

: 

+Rabatt       ==>^RAB@% 

)INIT 

   &NAME=&Z 

   &STR=&Z 

   &PLZ=&Z 

   &ORT=&Z 

   &RAB=&Z 

   .CURSOR=&NAME 

)REINIT 

   .ATTR(.CURSOR)='INTENS(HIGH) COLOR(RED)' 

)PROC 

   VER(&NAME,NB,MSG=MSG001) 

   VER(&VNAM,NB,MSG=MSG001)  

   VER(&VNAM,LEN,'<=',2,MSG=MSG001) 

   VER(&STR,NB,MSG=MSG001) 

   VER(&PLZ,NB,NUM,MSG=MSG002) 

   VER(&PLZ,RANGE,0,99999,MSG=MSG003) 

   IF (&PLZ = '85774') 

      &ORT = 'Unterföhring' 

      REFRESH(ORT) 

   VER(&ORT,NB,MSG=MSG001) 

   VER(&RAB,LIST,0,5,10,20,MSG=MSG004) 

)END 

 

4.10.9.7. Ablaufsteuerung 

Durch die Fragestellung mit IF kann der Inhalt einer Variablen überprüft und ein vom Prüfergebnis 
abhängiger JA- oder NEIN-Zweig durchlaufen werden. 

Die Kodierung des NEIN-Zweiges ist nicht zwingend, sofern von der Ablauflogik her im NEIN-Fall 
keine gesonderte Verarbeitung durchgeführt werden soll. 

Die Kodierung der IF-Klausel ist spaltensensitiv. Befehle, die an die erfüllte Bedingung des IF ge-
knüpft sind müssen eingerückt und spaltengleich geschrieben werden. Sobald eine Zeile spalten-
gleich zum IF oder links davon kodiert wird, wird der IF-Zweig als beendet betrachtet. Wird ein NEIN-
Zweig kodiert , ist er in der gleichen Spalte wie IF mit ELSE einzuleiten. Die abhängigen Befehle 
sind wiederum einzurücken und spaltengleich zu schreiben. 

Als Vergleichsoperanden in der Abfrage stehen zur Verfügung: 

Operator Aussage 

=, EQ Gleich 

^=, NE Ungleich 

>, GT Größer als 

>=, GE Größer oder gleich als 

^>, NG Nicht größer als 

<, LT Kleiner als 

<=, LE Kleiner oder gleich zu 

^<, NL Nicht kleiner als 

4.10.9.8. Feld-Aktualisierung 

Ändert sich der Wert einer Variablen innerhalb des )PROC-Teils oder im )REINIT-Bereich des Pa-
nels, ist ein REFRESH nötig, damit der geänderte Wert der Variablen bei erneuter Panelanzeige 
registriert wird. Die Aktualisierung (REFRESHING) von Variablen kann gezielt auf eine oder mehrere 
Variable hin, oder pauschal für alle Variable durchgeführt werden: 

REFRESH(var-1 [,var-n]) 

REFRESH(*) 

Angenommen, ein Feld &MWST soll geprüft werden. Abhängig vom Ergebnis wird ein Feld auf "IN-
LAND" oder "EXPORT" gesetzt. 

)PROC 

   VER(&MWST,LIST,0,19,MSG=MSG005) 

   IF (&MWST = 0) 

      &UMSART = 'EXPORT' 

   ELSE 

      &UMSART = 'INLAND' 

   REFRESH(UMSART) 

4.10.9.9. Zeichenfolgen abschneiden 

Über TRUNCATE kann in Verbindung mit einer Wertzuweisung eine Zeichenfolge nach bestimmten 
Kriterien zerlegt werden. Das Zerlegen erfolgt, indem die Zeichenfolge auf ein bestimmtes Zeichen 
durchsucht wird. Wenn gefunden, wird alles bis hin vor das Trennzeichen einer Variablen zugeord-
net. Der Rest wird in .TRAIL abgelegt. Das Trennzeichen geht verloren. Anderenfalls wird der ge-
samte Inhalt der Variablen übernommen. 

Die zweite Möglichkeit besteht in der Angabe einer Byteposition. Hier wird von Byte 1 bis zur ange-
gebenen Stelle eine Variable gefüllt, der Rest steht auch hier in der Steuervariablen .TRAIL zur 
Verfügung. Es geht nichts verloren. 

Das Beispiel zeigt das Zerlegen eines Datumsfeldes anhand der enthaltenen Punkte: 

)PROC 

   VER(&DATUM,NB,PICT,'NN.NN.NNNN') 

   &TT=TRUNC(&DATUM,'.') 

   &REST=.TRAIL 

   &MM=TRUNC(&REST,'.') 

   &JJ=.TRAIL 

Soll die im vorangegangenen Beispiel erzeugte Variable &MM inhaltlich überprüft und aufgrund des 
Ergebnisses eine Variable &MAXTAG mit den maximal möglichen Tagen abhängig zum Monat ge-
füllt werden, wäre im Grunde eine Mehrfachabfrage (vergleichbar zu SELECT in REXX) erforderlich. 
Da diese Mehrfachabfrage nicht existiert, muss anstelle dessen ein in sich geschachteltes IF/ELSE-
Konstrukt treten. Das nachfolgende Beispiel zeigt ein solches Gebilde: 

VER(&MM,RANGE,01,12) 

IF (&MM = 02) 

   &MAXTAG = 29 

ELSE 

   IF (&MM = 04,06,09,11) 

      &MAXTAG = 30 

   ELSE 

      &MAXTAG = 31 

Für eine sinnvolle Prüfung dieser Art, die auch ein mögliches Schaltjahr berücksichtigen kann, wäre 
allerdings die Einbettung von REXX-Funktionalität im Panel nötig, weil innerhalb des Panels keine 
Arithmetik stattfinden kann (wird später erklärt). 

4.10.9.10. Zeichenfolgen übersetzen  

Über die Funktion TRANSLATE kann in einer Wertzuweisung bei Erfüllung einer bestimmten Bedin-
gung der zuzuweisende Wert bestimmt werden. 

Auch hier soll vom Beispiel ausgehend eine Übertragung der Kundengruppe in die Variable 
&KUGRU abhängig vom Rabattsatz erfolgen. Wurde kein Rabatt gewährt, bleibt die Kundengruppe 
unverändert. 

)PROC 

   VER(&RAB,LIST,0,5,10,20,MSG=MSG003) 

   &KUGRU = TRANS(&RAB, 

          5,'Einzelhändler' 

         10,'Großhändler-A' 

         20,'Großhändler-B' 

           *,&KUGRU) 

)END 

4.10.9.11. Steuervariablen 

Panel-Steuervariablen weisen als Erkennungsmerkmal im Unterschied zu den Programm- oder Sys-
temvariablen einen Punkt auf und werden genutzt, um bestimmte generelle Aussagen zu treffen. 

Die wichtigsten Panel-Steuervariable in alphabetischer Reihenfolge und ihre mögliche Schreibweise 
sind: 

Steuervariable Bedeutung 

.ALARM Ausgabe Warnton 

.ATTR Temporäre Änderung eines Feldattributes 

.ATTRCHAR Änderung eines eingestellten Panel-Attributzeichens 

.CURSOR Cursorposition auf ein Feld 

.CRSPOS Cursorpositionierung innerhalb eines Feldes 

.HELP Zuweisung eines HELP-Panels 

.MSG Zeigen auf eine definierte Nachricht 

.RESP Ende der )BODY-Section durch ENTER,  END oder RE-
TURN 

.TRAIL Restinhalt einer Variablen nach TRUNCate-Funktion 

.ZVARS Reihenfolge der Platzhalter &Z im )BODY 

 

Die folgenden Beispiele verdeutlichen den Umgang mit den Panel-Steuervariablen: 

4.10.9.11.1. .ZVARS 

Werden im )BODY-Teil Variable benötigt, deren inhaltliche Länge kürzer als ihre Variablennamen 
sein soll, kann als Platzhalter die Variable &Z  für die betroffenen Felder benutzt werden. 

)BODY 

+Name        ==>_NAME                     + 

+Vorname     ==>_VNAM                      + 

+Alter       ==>_Z + 

+Geschlecht  ==>_Z+(M/W) 

+Beruf       ==>_Beruf            + 

+Studium     ==>_Z+(J/N) 

)INIT 

   .ZVARS='(ALTER,MANNWEIB,STUDIUM)' 

Im )INIT-Teil wird über die Steuervariable .ZVARS festgelegt, wie die Verbindung zwischen den 
Platzhaltern und den jeweiligen Variablen herzustellen ist. Die Platzhalter werden dabei im )BODY-
Teil von links nach rechts und von oben nach unten gelesen und ihre Inhalte in dieser Reihenfolge 
den Variablen zugewiesen. 

4.10.9.11.2. .ALARM .ATTR .ATTRCHAR .CSRPOS .MSG .RESP 

Diese Steuervariablen werden häufig im Verbund in Verbindung mit Fehlerbehandlung genutzt. 

)REINIT 

   .ATTRCHAR(%)='INTENS(LOW)' 

   .ATTRCHARS(_)='INTENS(LOW)' 

   .ATTR(.CURSOR)='INTENS(HIGH) 

)PROC 

   IF (.RESP ^= 'ENTER') 

      .MSG=MSG003 

      .ALARM=YES 

   VER(&DATUM,NB) 

   &TT=TRUNC(&DATUM,'.') 

   &REST=.TRAIL 

   &MM=TRUNC(&REST,'.') 

   &JJ=.TRAIL 

   VER(&TT,NB,RANGE,1,31) .CSRPOS = 1 

   VER(&MM,NB,RANGE,1,12) .CSRPOS = 4 

   VER(&JJ,NB,RANGE,1950,2040) .CSRPOS = 7 

)END 

Werden im )PROC-Teil Fehler erkannt, können auf dem Rückweg zum )BODY im )REINIT-Teil At-
tribute temporär geändert werden, um das fehlerhafte Feld bei neuer Präsentation hervorzuheben. 

4.10.10. Action Bars und Pull-Down-Menüs 

Der Umgang mit Actionbars polarisiert die Entwickler. Die einen lieben es und können nicht ohne 
leben, die anderen hassen es und verweigern sich. Ich selbst tendiere eher zur zweiten Gruppe. 
Dennoch soll hier auf die Möglichkeiten eingegangen werden. 

Eine Variante, Auswahlmöglichkeiten in einem Panel zu treffen, ist der Einsatz von Action Bars und 
Pull-Down-Menüs. Oben am Panelrand wird ein "Balken" eingeblendet, in dem verschiedene aus-
führbare Aktionen stehen. Wird der Cursor auf ein Feld des Aktionsblocks gesetzt und ENTER ge-
drückt, erscheinen die jeweiligen Wahlmöglichkeiten in Form eines nach unten gezogenen Menüs 
(Pull-Down), in dem durch Cursorpositionierung oder durch Zahleneingabe selektiert werden kann. 

Hier soll nicht auf die Details der ActionBar/Pulldown-Panel-Programmierung eingegangen werden, 
da dies den Rahmen sprengen würde. Dennoch werden die wichtigsten Elemente im Anschluss 
beschrieben. Ich muss mich an dieser Stelle auch outen. Ich habe eine tiefe Abneigung gegen Ac-
tionbars und PullDown Menüs – sowohl in der Kodierung als auch in der Bedienung. Im Übrigen 
halte ich Actionbars für einen Fehltritt aus den 1990er Jahren, als man unbedingt der Meinung war, 
man müsse etwas PC-Feeling auf den Mainframe bringen und lehne Actionbars in meinen eigenen 
Anwendungen ab. Kann natürlich auch daran liegen, dass ich ein "ewig Gestriger" bin, der sich ge-
gen neue Techniken sträubt. Wie dem auch sein, hier ein paar grundlegende Informationen zum 
Thema 

4.10.10.1. Action Bars 

Action Bar ist die "Wortleiste", die am oberen Panelrand, welche die Namen von Auswahlgruppen 
enthält (wer kennt das nicht aus Word mit <Datei><Start><Einfügen> und so weiter). 

4.10.10.2. Pull-Downs 

Pull-Downs erscheinen nach anwählen einer Auswahlgruppe der Action Bar. Die sichtbaren Mög-
lichkeiten beziehen sich auf die angewählte Gruppe. Nach Auswahl (Cursor oder Tastatureingabe) 
kann bereits im Panel eine Aktion erfolgen, oder die Funktion im Hintergrund entscheidet, was zu 
tun ist. 

Im Folgenden sind die dafür nötigen Bereiche beschrieben. Das Panel muss um die )ABC-Sektionen 
erweitert werden. 

4.10.10.3. )ABC (Action Bar Choice Section) 

Im )ABC-Teil wird ein Eintrag der Action Bar und die Untergruppen (Pull-Downs) definiert. Alle Ein-
träge bilden die Action Bar. Sie wird von links nach rechts gezeigt, wie von oben nach unten kodiert. 

4.10.10.4. )ABCINIT (Action Bar Initialization Section) 

Der )ABCINIT-Teil wird durchlaufen, wenn ein Feld der Action Bar angesteuert wird. Danach wird 
das Pull-Down-Menü gezeigt. 

4.10.10.5. )ABCPROC (Action Bar Processing Section) 

Der )ABC-PROC-Teil ist wahlfrei. Wird er kodiert, muss er aber mindestens ein Statement enthalten. 
Der )ABCPROC-Teil wird durchlaufen, wenn eine Auswahl in einem Pull-Down-Menü getroffen 
wurde. 

4.10.10.6. Die Reihenfolge 

1. )ABC 

2. )ABCINIT 

3. )ABCPROC 

muss eingehalten werden. 

4.10.10.7. Panelzusätze für Action Bars 

Eine Action Bar benötigt zusätzliche Attributtypen, die im )ATTR-Teil des Panels hinterlegt werden. 
Die wichtigsten Attribute sind: 

4.10.10.7.1. TYPE(AB) Action Bar (Auswahlmöglichkeit) 

Markiert eine Auswahlgruppe innerhalb der Auswahlleiste. Jeder Auswahlblock muss je aufs Neue 
mit dem Attribut TYPE(AB) gekennzeichnet werden. Zwischen Atrributszeichen und Text wird zwin-
gend ein BLANK erwartet. 

4.10.10.7.2. TYPE(NT) Normal Text 

Erlaubt vergleichbar zu den herkömmlichen Textattributen die Definition von Farben und anderen 
Charakteristiken. 

4.10.10.7.3. TYPE(ABSL)  Action Bar Separator Line 

Erzeugt eine Trennzeile zwischen Auswahlleiste und der restlichen Bildschirmmaske. 

4.10.10.7.4. TYPE(PT) Panel Title 

Liefert eine Panel-Überschrift. 

Hinweis 
Zwingend ist ausschließlich die Kodierung des Typs(AB). Die restlichen Attribute können prinzipiell 
herkömmlich kodiert werden (TYPE(TEXT) COLOR(GREEN) INTENS(LOW)....). Details hierzu sind 
auch in der IBM-Literatur unter den Schlagworten Panels und CUA-Mode zu finden. 

Beispiel 

Das folgende Beispiel zeigt ein Action Bar Panel. Zur Erläuterung wird die Auswahlgruppe SORT 
beschrieben. 

)ATTR FORMAT(MIX) 

   + TYPE(TEXT) 

   ! TYPE(AB) 

   # TYPE(NT) 

   [ TYPE(ABSL) 

)ABC 

   desc('Sort') 

   pdc desc('Abteilung') 

   pdc desc('Name') 

   pdc desc('Ort') 

)ABCINIT 

   .zvars=pdw 

   &pdw=&z 

)ABCPROC 

   ver(&pdw,LIST,1,2,3,4,5) 

   &rxprog='SORT' 

)BODY expand($$) 

+#! Wartung ! Sort ! Druck ! Ende# 

[$-$ Trennzeile AB - Panel $-$ 

+$-$< Telefonbuch Grundmenu >$-$ 

+Command ===>_zcmd 

+ 

)INIT 

   &rxprog=&z  

   &pdw   =&z 

)PROC 

)END 

Der Funktion sieht man nicht an, dass die Entscheidungen mit Hilfe von Action Bar und Pull-Downs 
getroffen wurden: 

/* REXX ******************************************/ 

Address ispexec 

do forever 

   "display panel(actbar)" 

   if rc=8 ! translate(rxprog)="ENDE" then leave 

   if length(rxprog) > 0 then do 

      interpret "call" rxprog 

      iterate 

   end 

   /* Verarbeitung ohne Auswahl über Action-Bar */ 

end 

exit 

sort: 

   select 

      when pdw=1 then sk="....." 

      when pdw=2 then sk="....." 

      when pdw=3 then sk="....." 

      otherwise nop 

   end 

   "tbsort telverz fields("sk")" 

Return 

 

Erweitertes Beispiel Action Bar 

(stammt aus dem IBM Knowledge Center)  

)PANEL 

Wird gebraucht, damit das Panel im CUA Modus laufen kann. 

)ATTR 

 @ TYPE(AB) 

 # TYPE(NT) 

⋮ 

Die Definitionen in der )ATTR Section legen fest, dass über @ ein Feldtyp "ACTION BAR" und über 
# ein Feldtyp "NORMAL TEXT" definiert wird. 

)ABC DESC(FILE) MNEM(1) 

  PDC DESC(file-choice1) ACC(Alt+F1) 

  ACTION RUN(command-name) PARM(command-parms) 

  PDC DESC(file-choice2) UNAVAIL(&unvar2) 

  ACTION RUN(command-name) PARM(command-parms) 

  PDC DESC(file-choice3) PDSEP(ON) 

  ACTION RUN(command-name) PARM(command-parms) 

Erster Eintrag in der Action Bar mit dem Namen FILE, öffnet, wenn angesprungen, ein PullDown 
Menü mit drei verschiedenen Auswahlmöglichkeiten. Abhängig von der jeweiligen Auswahl be-
schreibt die ACTION RUN Anweisung, was jeweils geschehen soll. 

)ABCINIT 

  .ZVARS = PDCHOICE 

  &PDCHOICE = '' 

  &unvar2 = 1 

⋮ 

Initialisierungssection für FILE 

)ABCPROC 

  VER (&PDCHOICE,LIST,1,2,3) 

⋮ 

Verarbeitungssection für FILE 

)ABC DESC(HELP) 

  PDC DESC(help-choice1) MNEM(6) 

  ACTION RUN(command-name) PARM(command-parms) 

  PDC DESC(help-choice2) 

  ACTION RUN(command-name) 

  PDC DESC(help-choice3) 

  ACTION RUN(command-name) PARM(command-parms) 

⋮ 

Zweiter Eintrag der Action Bar mit dem Namen HELP. Auch hier werden die diversen Möglichkeiten 
des entsprechenden PullDown Menüs und deren abhängiger Aktionen definiert. 

)ABCINIT 

  .ZVARS = PDCHOICE 

  &PDCHOICE = ' 

⋮ 

Initialisierungssection für HELP 

)ABCPROC 

  VER (&PDCHOICE,LIST,1,2,3) 

⋮ 

Verarbeitungssection für HELP 

)BODY 

 @ FILE@ HELP# 

⋮ 
)END 

In wie weit man sich das antun muss oder will oder ob man es auf Grund von Vorgaben einfach tun 
MUSS, ist eine andere Sache. Die Kodierung von Action Bars kann durchaus sehr umfangreich 
werden. Es gehen einem auch schnell die Sonderzeichen für Attribute auf der Tastatur aus, weshalb 
dann oft auf HEX-Attributszeichen ausgewichen wird, was die Lesbarkeit im Panel nicht verbessert). 
Im Folgenden ein unkommentiertes Beispiel des ISR@PRIM (Startpanel für ISPF/PDF) bis zur 
)BODY Section (ist kompliziert genug). Ohne Action Bars könnte das Panel mit wenigen Zeilen ko-
diert werden: 

)PANEL KEYLIST(ISRSAB,ISR) IMAGE(&ZIMGNAM,&ZIMGROW,&ZIMGCOL) 

)ATTR DEFAULT(   ) FORMAT(MIX)            /* ISR§PRIM - ENGLISH - 7.4 */ 

 0B TYPE(AB) 

 0D TYPE(PS) 

 04 TYPE(ABSL) GE(ON) 

 05 TYPE(PT) 

 09 TYPE(FP) 

 0A TYPE(NT) 

 11 TYPE(SAC) 

 16 TYPE(VOI) PADC(USER) 

 2B TYPE(DT) 

 22 TYPE(WASL) SKIP(ON) GE(ON) 

 26 TYPE(NEF) CAPS(ON) PADC(USER) 

 27 AREA(SCRL) EXTEND(ON) 

 28 TYPE(PS) CSRGRP(99) 

 29 TYPE(GRPBOX) 

 2A AREA(DYNAMIC) 

 19 TYPE(DATAOUT) COLOR(WHITE) 

 1A TYPE(DATAOUT) COLOR(RED) 

 1B TYPE(DATAOUT) COLOR(BLUE) 

 1C TYPE(DATAOUT) COLOR(GREEN) 

 1D TYPE(DATAOUT) COLOR(PINK) 

 1E TYPE(DATAOUT) COLOR(YELLOW) 

 1F TYPE(DATAOUT) COLOR(TURQ) 

 30 TYPE(DATAOUT) PAS(ON) 

 31 TYPE(CHAR) COLOR(WHITE) 

 32 TYPE(CHAR) COLOR(RED) 

 33 TYPE(CHAR) COLOR(BLUE) 

 34 TYPE(CHAR) COLOR(GREEN) 

 35 TYPE(CHAR) COLOR(PINK) 

 36 TYPE(CHAR) COLOR(YELLOW) 

 37 TYPE(CHAR) COLOR(TURQ) 

 38 TYPE(DATAOUT) CUADYN(VOI) 

 39 TYPE(CHAR) COLOR(WHITE) HILITE(REVERSE) 

 3A TYPE(CHAR) COLOR(RED) HILITE(REVERSE) 

 3B TYPE(CHAR) COLOR(BLUE) HILITE(REVERSE) 

 3C TYPE(CHAR) COLOR(GREEN) HILITE(REVERSE) 

 3D TYPE(CHAR) COLOR(PINK) HILITE(REVERSE) 

 3E TYPE(CHAR) COLOR(YELLOW) HILITE(REVERSE) 

 3F TYPE(CHAR) COLOR(TURQ) HILITE(REVERSE) 

)ABC DESC('Menu') MNEM(1) 

PDC DESC('Settings') UNAVAIL(ZPM1) MNEM(1) ACC(CTRL+S) 

 ACTION RUN(ISRROUTE) PARM('SET') 

PDC DESC('View') UNAVAIL(ZPM2) MNEM(1) ACC(CTRL+V) 

 ACTION RUN(ISRROUTE) PARM('BR1') 

PDC DESC('Edit') UNAVAIL(ZPM3) MNEM(1) ACC(CTRL+E) 

 ACTION RUN(ISRROUTE) PARM('ED1') 

PDC DESC('ISPF Command Shell') UNAVAIL(ZPM4) MNEM(6) ACC(CTRL+C) 

 ACTION RUN(ISRROUTE) PARM('C1') 

PDC DESC('Dialog Test...') UNAVAIL(ZPM5) MNEM(8) ACC(CTRL+T) 

 ACTION RUN(ISRROUTE) PARM('DAL') 

PDC DESC('Other IBM Products...') UNAVAIL(ZPM6) MNEM(1) ACC(CTRL+O) 

 ACTION RUN(ISRROUTE) PARM('OIB') 

PDC DESC('SCLM') UNAVAIL(ZPM7) MNEM(3) ACC(CTRL+L) 

 ACTION RUN(ISRROUTE) PARM('SCL') 

PDC DESC('ISPF Workplace') UNAVAIL(ZPM8) MNEM(6) ACC(CTRL+W) 

 ACTION RUN(ISRROUTE) PARM('WRK') 

PDC DESC('Status Area...') UNAVAIL(ZPMS) MNEM(8) ACC(CTRL+A) 

 ACTION RUN(ISRROUTE) PARM('SAM') 

PDC DESC('Exit') MNEM(2) PDSEP(ON) ACC(CTRL+X) ACTION RUN(EXIT) 

)ABCINIT 

.ZVARS=ISR§OPT 

)ABC DESC('Utilities') MNEM(1) 

PDC DESC('Library') UNAVAIL(ZUT1) MNEM(1) ACC(ALT+1) 

 ACTION RUN(ISRROUTE) PARM('U1') 

PDC DESC('Data set') UNAVAIL(ZUT2) MNEM(1) ACC(ALT+2) 

 ACTION RUN(ISRROUTE) PARM('U2') 

PDC DESC('Move/Copy') UNAVAIL(ZUT3) MNEM(1) ACC(ALT+3) 

 ACTION RUN(ISRROUTE) PARM('U3') 

PDC DESC('Data Set List') UNAVAIL(ZUT4) MNEM(2) ACC(ALT+4) 

 ACTION RUN(ISRROUTE) PARM('U4') 

PDC DESC('Reset Statistics') UNAVAIL(ZUT5) MNEM(5) ACC(ALT+5) 

 ACTION RUN(ISRROUTE) PARM('U5') 

PDC DESC('Hardcopy') UNAVAIL(ZUT6) MNEM(1) ACC(ALT+6) 

 ACTION RUN(ISRROUTE) PARM('U6') 

PDC DESC('Download...') UNAVAIL(ZUTDT) MNEM(2) ACC(ALT+7) 

 ACTION RUN(ISRROUTE) PARM('UDT') 

PDC DESC('Outlist') UNAVAIL(ZUT7) MNEM(2) ACC(ALT+8) 

 ACTION RUN(ISRROUTE) PARM('U8') 

PDC DESC('Commands...') UNAVAIL(ZUT8) MNEM(1) ACC(ALT+9) 

 ACTION RUN(ISRROUTE) PARM('U9') 

PDC DESC('Reserved') UNAVAIL(ZUT9) MNEM(6) ACTION RUN(ISRROUTE) PARM('U10') 

PDC DESC('Format') UNAVAIL(ZUT10) MNEM(1) ACC(ALT+F1) 

 ACTION RUN(ISRROUTE) PARM('U11') 

PDC DESC('SuperC') UNAVAIL(ZUT11) MNEM(1) PDSEP(ON) ACC(CTRL+F2) 

 ACTION RUN(ISRROUTE) PARM('U12') 

PDC DESC('SuperCE') UNAVAIL(ZUT12) MNEM(3) ACC(CTRL+F3) 

 ACTION RUN(ISRROUTE) PARM('U13') 

PDC DESC('Search-For') UNAVAIL(ZUT13) MNEM(2) ACC(CTRL+F4) 

 ACTION RUN(ISRROUTE) PARM('U14') 

PDC DESC('Search-ForE') UNAVAIL(ZUT14) MNEM(4) ACC(CTRL+F5) 

 ACTION RUN(ISRROUTE) PARM('U15') 

PDC DESC('Table Utility') UNAVAIL(ZUT15) MNEM(3) ACC(CTRL+F6) 

 ACTION RUN(ISRROUTE) PARM('U16') 

PDC DESC('Directory List') UNAVAIL(ZUT16) MNEM(2) ACC(CTRL+F7) 

 ACTION RUN(ISRROUTE) PARM('U17') 

)ABCINIT 

.ZVARS=PDFUTIL 

     &zut9 = '1' 

)ABC DESC('Compilers') MNEM(1) 

PDC DESC('Foreground Compilers') MNEM(1) ACTION RUN(ISRROUTE) PARM('FGD') 

PDC DESC('Background Compilers') MNEM(1) ACTION RUN(ISRROUTE) PARM('BKG') 

PDC DESC('ISPPREP Panel Utility...') MNEM(1) ACTION RUN(ISPPREP) 

PDC DESC('DTL Compiler...') MNEM(1) ACTION RUN(ISPDTLC) 

)ABCINIT 

.ZVARS=ISRLANG 

)ABC DESC('Options') MNEM(1) 

PDC DESC('General Settings') MNEM(1) ACTION RUN(ISRROUTE) PARM('SET') 

PDC DESC('CUA Attributes...') MNEM(1) ACTION RUN(CUAATTR) 

PDC DESC('Keylists...') MNEM(1) ACTION RUN(KEYLIST) 

PDC DESC('Point-and-Shoot...') MNEM(1) ACTION RUN(PSCOLOR) 

PDC DESC('Colors...') MNEM(2) ACTION RUN(COLOR) 

PDC DESC('Dialog Test appl ID...') MNEM(1) ACTION RUN(ISRROUTE) PARM('DAP') 

)ABCINIT 

.ZVARS=PDFOPTM 

)ABC DESC('Status') MNEM(1) 

PDC DESC('Session') UNAVAIL(ZCSTA) MNEM(1) ACTION RUN(ISRROUTE) PARM('SES') 

PDC DESC('Function keys') UNAVAIL(ZCSTB) MNEM(1) 

 ACTION RUN(ISRROUTE) PARM('FUN') 

PDC DESC('Calendar') UNAVAIL(ZCSTC) MNEM(1) ACTION RUN(ISRROUTE) PARM('CAL') 

PDC DESC('User status') UNAVAIL(ZCSTD) MNEM(1) ACTION RUN(ISRROUTE) PARM('USE')  

PDC DESC('User point and shoot') UNAVAIL(ZCSTE) MNEM(3) 

 ACTION RUN(ISRROUTE) PARM('UPS') 

PDC DESC('None') UNAVAIL(ZCSTF) MNEM(1) ACTION RUN(ISRROUTE) PARM('OFF') 

)ABCINIT 

.ZVARS=PDFSTA 

IF (&ZSCREEN = '1') 

  VGET (ZSAREA1) 

  &ZSAREA = &ZSAREA1 

  GOTO A 

IF (&ZSCREEN = '2') 

  VGET (ZSAREA2) 

  &ZSAREA = &ZSAREA2 

  GOTO A 

IF (&ZSCREEN = '3') 

  VGET (ZSAREA3) 

  &ZSAREA = &ZSAREA3 

  GOTO A 

IF (&ZSCREEN = '4') 

  VGET (ZSAREA4) 

  &ZSAREA = &ZSAREA4 

  GOTO A 

ELSE 

  VGET (ZSAREA5) 

  &ZSAREA = &ZSAREA5 

A: 

&zcsta = 0 

&zcstb = 0 

&zcstc = 0 

&zcstd = 0 

&zcste = 0 

&zcstf = 0 

IF (&ZSAREA = 'SES') &zcsta = 1 

IF (&ZSAREA = 'FUN') &zcstb = 1 

IF (&ZSAREA = 'CAL') &zcstc = 1 

IF (&ZSAREA = 'USE') &zcstd = 1 

IF (&ZSAREA = 'UPS') &zcste = 1 

IF (&ZSAREA = 'OFF') &zcstf = 1 

&PDFSTA = ' ' 

)ABC DESC('Help') MNEM(1) 

PDC DESC('General') MNEM(1) ACTION RUN(TUTOR) PARM('ISR01000') 

PDC DESC('Settings') MNEM(1) ACTION RUN(TUTOR) PARM('ISP05000') 

PDC DESC('View') MNEM(1) ACTION RUN(TUTOR) PARM('ISR10000') 

PDC DESC('Edit') MNEM(1) ACTION RUN(TUTOR) PARM('ISR20000') 

PDC DESC('Utilities') MNEM(1) ACTION RUN(TUTOR) PARM('ISR30000') 

PDC DESC('Foreground') MNEM(1) ACTION RUN(TUTOR) PARM('ISR40000') 

PDC DESC('Batch') MNEM(1) ACTION RUN(TUTOR) PARM('ISR50000') 

PDC DESC('Command') MNEM(1) ACTION RUN(TUTOR) PARM('ISR60010') 

PDC DESC('Dialog Test') MNEM(1) ACTION RUN(TUTOR) PARM('ISP70000') 

PDC DESC('IBM Products') MNEM(1) ACTION RUN(TUTOR) PARM('ISRD0000') 

PDC DESC('SCLM') MNEM(3) ACTION RUN(TUTOR) PARM('FLMTD') 

PDC DESC('Workplace') MNEM(1) ACTION RUN(TUTOR) PARM('ISR00400') 

PDC DESC('Exit') MNEM(2) ACTION RUN(TUTOR) PARM('ISP90100') 

PDC DESC('Status Area') MNEM(2) ACTION RUN(TUTOR) PARM('ISPSAMHP') 

PDC DESC('About...') MNEM(1) ACTION RUN(TUTOR) PARM('ISR§LOGO') 

PDC DESC('Changes for this Release') MNEM(2) ACTION RUN(TUTOR) PARM('ISR00005') 

PDC DESC('Tutorial') MNEM(4) ACTION RUN(TUTOR) PARM('ISR00000') 

PDC DESC('Appendices') MNEM(2) ACTION RUN(TUTOR) PARM('ISR00004') 

PDC DESC('Index') MNEM(2) ACTION RUN(TUTOR) PARM('ISR91000') 

)ABCINIT 

.ZVARS=PRIMHELP 

)BODY  CMD(ZCMD) 

 Menu   Utilities Compilers  Options   Status   Help 

 ------------------------------------------------------------------------  

                          ISPF Primary Option Menu 

4.11. Selection-Panels (Auswahl-Bildschirme) 

Der Unterschied zu Datenpanels besteht darin, dass hier normalerweise nicht mehrere Felder zur 
Dateneingabe präsentiert werden, sondern unter mehreren angebotenen Funktionen durch eine Ein-
gabe entschieden wird, welche Aktion im Anschluss stattfinden soll (siehe Grundmenü des PDF). 

Als mögliche Aktionen können Aufrufe für  

▪ PANELS, die direkt aufgerufen werden, ohne die Steuerung an die Funktion zurückgeben zu 
müssen 

▪ Loadmodules (Ablauffähige Programme), die innerhalb der ISPLLIB-Zuordnung zu finden 
sein müssen 

▪ Prozeduren (CLIST oder REXX), die unter der Zuordnung von SYSPROC/SYSEXEC ge-
sucht werden 

stattfinden. 

Im Auswahlpanel können die gleichen Sections wie in einem Datenpanel vorkommen. )BODY und 
)PROC sind zwingend. Durch die  Angabe  "&ZPRIM=YES" kann eine Auswahlmaske als so ge-
nanntes Primary-Option-Menü definiert werden. Die Folge daraus ist 

▪ bei Benutzung der RETURN-Funktion wird in einem hierarchischen Dialog hier Halt gemacht 
▪ Sprungbefehle wie "=3.2" können nur unterhalb dieser Hierarchiestufe benutzt werden. 
▪ Das direkte Verlassen eines Dialoges durch "=X" bewirkt einen Rücksprung in die nächste 

hierarchisch übergeordnete Ebene. 

4.11.1. Die )BODY-Section im Selection-Panel 

Für die Kodierung gelten folgende Regeln: 

▪ Es muss mindestens ein Eingabefeld für die Aufnahme der ausgewählten Option vorhanden 
sein. Im Allgemeinen wird hierfür das ZCMD-Feld benutzt, es kann aber auch das erste Feld 
nach dem ZCMD-Feld sein. 

▪ Weitere Eingabefelder können vorhanden sein. In der Regel ist aber im Hinblick auf eine 
übersichtliche Bedienerführung möglichst darauf zu verzichten. 

▪ Das Selection-Panel hat keinen Zugriff auf den Funktion-Pool. Es können nur Variable aus 
dem Shared-Pool oder Profile-Pool abgebildet werden. 

4.11.2. Die )PROC-Section im Selection-Panel 

Besonderheiten sind: 

▪ Der &ZSEL-Service muss im )PROC-Teil kodiert sein. Hier erfolgt über eine eingebettete 
SELECT-Funktion die Entscheidung, welche Aktivität aufgrund der Eingabe eingeleitet wer-
den soll. 

▪ Wird eine gegliederte Auswahl getroffen wie beispielsweise 3.5, sorgt der  Select Service für 
die entsprechende Weitergabe. Hierbei werden zu überspringende Panels verarbeitet, aber 
nicht angezeigt. 

Als Beispiel für ein Auswahlpanel betrachten wir ein mögliches Primary-Option-Menü des ISPF/PDF: 

)BODY EXPAND(//) 

%/*/ ISPF/PDF PRIMARY OPTION MENU /*/ 

%OPTION ==>_ZCMD                                      + 

% 

% 0 + ISPF PARMS - Specify terminal 

% 1 + BROWSE    - Display source data ......... 

% 2 + EDIT      - Create or change ........... 

% 3 + UTILITIES - Perform utility functions ... 

: 

% T + TUTORIAL 

% 

% PH+ PHONE     - INTERNAL PHONEBOOK 

)INIT 

   .HELP=ISP00003 

   &ZPRIM=YES 

   &ZHTOP=ISP00003 

)PROC 

   &ZSEL=TRANS(TRUNC(&ZCMD,'.') 

         0,'PANEL(ISPOPTA)' 

         1,'PGM(ISRBRO) PARM(ISRBRO01)' 

         2,'PGM(ISREDIT) PARM(P,ISREDM01)' 

         3,'PANEL(ISRUTIL)' 

         : 

         T,PGM(ISPTUTOR) PARM(ISP00000)' 

         : 

         PH,'CMD(PHONBOOK)' 

        : 

4.12. Hierarchische Dialogstrukturen 

Um die hierarchischen Strukturen eines Dialoges aufzuzeigen, betrachten wir uns PDF in graphi-
scher Darstellung. Das hierarchisch oberste Panel ist das PRIMARY-OPTION-MENU. Es wird durch 
einen SELECT PANEL Service beim Start des PDF aufgerufen. 

Untergeordnete Panels, die aus einem Programm aufgerufen werden, sind Datenpanels. Sie stellen 
das Ende einer Hierarchie dar und werden über den DISPLAY PANEL Service aktiviert. 

4.13. Messages 

Messages sind Nachrichten die situationsabhängig in den vorgesehenen Feldern SMSG (Short Mes-
sage) oder LMSG (Long Message) präsentiert werden können. 

Grundsätzlich unterscheiden wir zwischen Nachrichten, die innerhalb eines Panels aufgerufen wer-
den und solchen, die über die steuernde Funktion (z.B.: REXX) abgehandelt werden. Die Verarbei-
tung in beiden Fällen ist jedoch die gleiche. 

Alle im Dialog Manager zu verarbeitenden Meldungen müssen innerhalb einer unter ISPMLIB zuge-
ordneten PO-Datei unter Berücksichtigung bestimmter Formalismen in beliebigen Membern unter-
gebracht sein. 

4.13.1. Meldungsdefinition 

Beim Verweis auf eine Meldung geht aus der Meldungsdefinition der Membername innerhalb der 
ISPMLIB-Datei und die darin liegende gewünschte Meldung hervor. 

 
 
Die maximale Länge ist 8 Byte. 

Das Suffixzeichen darf nur benutzt werden, sofern nicht bereits 5 Prefixzeichen benutzt wurden. 

Der Membername, in welcher die Meldung innerhalb ISPMLIB gesucht wird, bildet sich aus dem 
Alpha-Prefix und den ersten beiden numerischen Werten. Eine Meldung AKUV001A muss also zwin-
gend im Member AKUV00 in ISPMLIB gefunden werden. 

4.13.2. Meldungsformat 

Die Datei, in welche die Meldungsmember eingetragen werden ist mit einer festen Satzlänge von 80 
Byte zu erstellen. Das Dateiprofil muss auf "NUMBER OFF" gesetzt sein. 

Jede Meldung besteht aus einem zweizeiligen Eintrag. Ab Spalte 1 der ersten Zeile steht die Mes-
sage-ID. Anschließend folgt der kurze Meldungstext zwischen Hochkommata. Die kurze Nachricht 
darf maximal 24 Byte (Ohne Hochkommata) lang sein. 

Der lange Meldungstext steht in der nächsten Zeile und ist ebenfalls zwischen Apostrophe einge-
schlossen. Die maximale Länge ist 78 Byte. 

In beiden Meldungstextzeilen können Variable innerhalb der Hochkommata genutzt werden. Die 
Variablen werden vor Anzeige der Meldung substituiert. 

Sinnvoll sind auch hier sprechende Namen. So ist empfehlenswert, die Message-ID und damit den 
Membernamen innerhalb von ISPMLIB mit der Applikation in Verbindung zu bringen, für die das 
Meldungspaket genutzt werden soll. Neutrale Namen können für Meldungen genutzt werden, die 
nicht applikationsspezifisch sind.  

4.13.3. Schlüsselwortparameter 

Die Aussagekraft der Messages kann mit verschiedenen Parametern verstärkt werden. Die Schlüs-
selwortparameter werden im Anschluss an die kurze Meldung kodiert. 

4.13.3.1. .HELP=hpname  

zeigt auf ein HELP-Panel, das präsentiert wird, wenn nach der langen Meldung die HELP-Taste 
gedrückt wird. 

4.13.3.2. .ALARM=YES|NO  

steuert, ob ein akustisches Signal bei Erscheinen der Meldung ausgelöst werden soll. 

4.13.3.3. .WINDOW=RESP|NORESP  

bewirkt, dass die Meldung in einem Fenster erscheint. RESP erzwingt das Drücken der ENTER-
Taste nachdem die Meldung erschienen ist, NORESP nicht. 

4.13.3.4. .TYPE=NOTIFY|WARNING|ACTION|CRITICAL  

verändert Farbe und Darstellung der Meldung je nach Wichtigkeit, in der folgenden Tabelle die Über-
sicht:  

Type Farbe Erscheint Enter zwingend Alarm 

Notify Weiß Normal oder als Fenster Nein Nein 

Warning Gelb Normal oder als Fenster Nein Nein 

Action Rot Normal oder als Fenster Nein Ja 

Critical Rot Immer als Fenster Nein Ja 

 

Beispiele 

FAKT002A 'Datumsformat falsch' .HELP=hpname 

'Das Datumsformat muss TT.MM.JJJJ sein!' 

Messagebeispiel mit HELP-Panel. Wird nach dem Erscheinen der langen Meldung die HELP-Taste 
gedrückt, wird das angegebene HELP-Panel nachgerufen. 

FAKT003A 'Wert falsch!' .WINDOW=NORESP .ALARM=YES 

'Der Wert muss zwischen 1 und 5 liegen.' 

Lässt die Nachricht in einem Fenster erscheinen (ohne Bestätigung mit ENTER) und den Warnton 
erklingen. 

4.13.4. Anzeige der Meldungen 

Sind keine benutzereigenen Meldungen definiert, werden vom Dialog Manager Standardmeldungen 
benutzt. 

4.13.5. Anzeige von HELP-Panels 

Generell können an HELP-Panels weitere HELP-Panels angeschlossen werden. Ist ein HELP-Panel 
sichtbar, kann der Benutzer des Dialoges durch drücken der ENTER-Taste von einem HELP-Panel 
zum Nächsten blättern. 

4.13.6. Ablauf der Meldungsanzeige 

Es ist ein Fehler aufgetreten. 

1. Sofern eine Meldung definiert ist, wird die Kurznachricht ausgegeben.  
2. Wird HELP gedrückt, wird die Langnachricht ausgegeben. 
3. Wird HELP gedrückt, wird das Help-Panel angezeigt 

4.13.7. Ausgabeort der Meldungen 

Soll vom Standard abgewichen werden (kurze Meldung in Zeile 1 rechts, lange Meldung in Zeile 3) 
sind entsprechende Definitionen im )BODY-Teil nötig. 

Im )ATTR-Teil muss für die Meldungsausgabe ein Attribut mit Type Output definiert sein, da es ein 
solches standardgemäß nicht gibt. 

)ATTR 

   $ TYPE(OUTPUT) INTENS(HIGH) 

)BODY EXPAND(//) SMSG(KMESS) LMSG(LMESS) 

+/-/ Überschrift /-/ 

+COMMAND ==>_ZCMD                                  + 

: 

:   Text-, Eingabe- und Ausgabefelder 

: 

$KMESS   + 

$LMESS                                             + 

Die Kurz Nachricht erscheint in der vorletzten Zeile, die lange in der letzten Zeile jeweils linksbündig. 

4.13.8. Steuerung der  Meldungs-Anzeige 

Die Anzeige der Meldungen kann entweder über Angaben in der )PROC-Section oder durch eine 
Dialog Manager Routine in einer Funktion (REXX) gesteuert werden. Wir sehen uns zu beiden Mög-
lichkeiten jeweils ein Beispiel an: 

)PROC 

   IF (&RAB='') 

      .MSG = FAKT001A 

   VER (&ORT,NB,MSG=FAKT002A) 

   : 

)END 

 

 

: 

DO FOREVER 

   "DISPLAY PANEL(pname)" 

   IF DATOK(datum)=1 THEN LEAVE 

   "SETMSG MSG(FAKT003A)" 

END 

: 

4.14. Tabellen 

Eine Tabelle ist ein im Allgemeinen zweidimensionales Gebilde aus Spalten und Zeilen, wobei spal-
tengleich sinngemäß gleiche Inhalte abgebildet sind. 

Alle zu einer Ebene (Zeile) zusammengefassten Daten ergeben einen  Datensatz. Die einzelnen 
Spalteneinträge innerhalb des Satzes bezeichnet man als Glied oder Feld oder Tabellenelement. 

Tabellen werden als Member von PO-Dateien geführt, die unabhängig von der Größe der Tabelle 
mit einer festen Satzlänge von 80 Byte angelegt werden. Die symbolischen Namen für das Lesen 
und Schreiben von Tabellen sind unterschiedlich. Das Lesen wird über den symbolischen Namen 
"ISPTLIB", das Schreiben über "ISPTABL" durchgeführt, wobei durchaus beide symbolischen Na-
men auf die gleiche PO-Datei zugewiesen sein können. 

Bei Verarbeitung einer Tabelle wird diese zunächst komplett in den virtuellen Speicher gelesen. 
Sämtliche Manipulationen an der Tabelle finden hier statt. Nach Veränderungen wird die gesamte 
Tabelle gegebenenfalls wieder auf Platte zurückgeschrieben. 

Die Größe einer Tabelle ist theoretisch beschränkt. In der Praxis stellen diese Grenzen aber im 
Allgemeinen keine Einschränkung dar. Die maximale Feldgröße (Variablengröße) einer Dialog Ma-
nager Tabelle liegt bei 32 Kb. Die maximale Gesamtgröße ist abhängig von der Größe des verfüg-
baren virtuellen Speichers. 

4.14.1. SAVE-Variablen – mehr Schatten als Licht 

Generell möchte ich hier zentral ein Hinweis auf die sogenannten SAVE-Variablen liefern. Es klingt 
phantastisch, wenn man für eine bestehende Tabelle beliebige Felder anhängen kann. Noch besser: 
man muss es nicht für jeden Satz tun. Unglaublich: die Namen der "Anhängsel" können bei jedem 
Satz unterschiedlich viele sein, voneinander beliebig abweichen und die Inhalte sind vollkommen 
wahlfrei. Aber: Vergesst es. Die Notwendigkeit so etwas zu machen bedeutet in der Regel, dass 
man beim Design der Tabelle zu schluderig war. 

Was wäre denn bei DB2 zu machen? Es gäbe einen ALTER COLUMN und "vergessene" Felder 
können hinzugefügt werden. Allerdings die gleichen an jeden Tabellensatz. Das halte ich für "nor-
mal". Der Umgang mit den SAVE-Variablen ist äußerst gewöhnungsbedürftig und bedarf einer auf-
wändigeren Programmlogik. Und warum sollte man sich das antun, wenn es nicht absolut notwendig 
ist? 

Wenn am Tabellendesign etwas zu ändern ist, geht man am besten den Weg: neue Tabelle mit der 
richtigen Struktur anlegen. Alte Tabelle auslesen und die neue befüllen. Alte Tabelle löschen. Neue 
Tabelle umbenennen. Klar, ist Arbeit, spart aber eine Menge Ärger und die Sourcecode-Anpassung 
für den Umgang mit den SAVE-Variablen ist keine wirkliche Freude. 

Für den fortgeschrittenen Anwender stellen die SAVE-Variablen natürlich ein, für den Bedarfsfall 
legitimes, Werkzeug dar. Die Member des Profile-Pools sind in der Regel ISPF-Tabellen ohne 
Schlüssel- und Namensfelder und bestehen ausschließlich aus SAVE-Variablen. Wer aber SO! weit 

vorgedrungen ist, wird auch damit kein Problem haben. Und wenn doch: einfach fragen    . Wir 

lassen diese Variante im weiteren Verlauf unberücksichtigt. 

4.14.2. Zugriff auf Tabellensätze 

Der Zugriff erfolgt immer auf eine komplette Zeile aus der Tabelle, unabhängig, ob eine sequentielle 
Verarbeitung durchgeführt wird, oder ob über SCAN-Vorgänge nach bestimmten Kriterien gefiltert 
werden soll. 

Der aktuelle Tabellensatz wird über einen so genannten CRP (Current Row Pointer oder Satzzeiger) 
markiert. Der CRP kann vorwärts und rückwärts positioniert werden. Nach dem Öffnen einer Tabelle 
steht der CRP auf "TOP" (am Anfang der Tabelle noch vor dem ersten Datensatz). 

Die Inhalte aller Tabellenfelder des Satzes, auf den der CRP zeigt sind unmittelbar über gleichna-
mige Variable im Function-Pool verfügbar. 

Sollen Tabelleneinträge verändert werden muss bei Tabellen ohne Schlüssel zunächst der CRP auf 
den zu ändernden Satz positioniert werden. Anschließend kann der Inhalt der feldnamensgleichen 
Variablen entsprechend verändert werden, bevor eine Aktualisierung der Tabelle mittels DMS-Be-
fehles erfolgt. Alle Änderungen sind zunächst nur temporär, da sie nur im virtuellen Speicher erfol-
gen. 

4.14.3. Tabellen erzeugen 

Um eine Tabelle erstellen zu können, muss man sich zunächst mit zwei Begriffen auseinanderset-
zen: 

4.14.3.1. KEYS 

Keys benennt die Schlüsselfelder. Prinzipiell kann eine Tabelle ohne Schlüsselfelder definiert wer-
den. Es können aber auch beliebig viele Felder zum Schlüssel erhoben werden.  

Sinn des Schlüssels ist, die Eintragung zweier Sätze, die in allen Schlüsselfeldern identisch sind, zu 
verhindern. 

4.14.3.2. NAMES 

beschreibt alle Tabellenfelder über die Schlüsselfelder hinaus. Soll eine Tabelle keine Schlüssel 
enthalten, müssen alle Felder innerhalb NAMES angegeben werden. 

4.14.3.3. Feldformate und -längen 

Angaben über Feldlängen und Feldformate sind nicht erforderlich. Die Felder werden in Bezug auf 
Länge und Format so in die Tabelle eingearbeitet, wie sie als Variable im Function-Pool  vorbereitet 
werden. 

4.14.4. Sortierung 

Häufig wird nach dem Erstellen einer Tabelle (sofern sie sortiert vorliegen soll) mit der Instruktion 
"TBSORT FIELDS(f1,x,y[...])" die Sort-Order innerhalb der Tabelle festgelegt. Über einen Operan-
den "ORDER" kann später auf die festgesetzte Sortierfolge Bezug genommen werden. 

4.14.5. Tabellen einlesen 

Über die Instruktion "TBOPEN tabname" wird ein Abbild der Tabelle im virtuellen Speicher erstellt. 
Durch entsprechende Parameter kann gesteuert werden: 

▪ Die Tabelle wird nur temporär geführt. Beim Schließen findet kein Rückschreiben auf Platte 
statt. 

▪ Die Tabelle wird beim Schließen auf Platte zurückgeschrieben. 

▪ Die Tabelle wird beim Rückschreiben unter einem anderen als dem vorgesehenen symboli-
schen Namen abgestellt (nicht empfehlenswert ohne Zwang). 

▪ Die Tabelle kann am Terminal im SPLIT-SCREEN-Modus verarbeitet werden. 

4.14.6. Tabelle schreiben 

Zunächst sind zwei unterschiedliche Befehle für das Rückschreiben einer Tabelle verfügbar. Mit 
TBSAVE kann ein Rückschreiben der Tabelle erfolgen, ohne diese schließen zu müssen. Mit 
"TBCLOSE tabname" erfolgt das Rückschreiben in Verbindung mit dem Schließen der Tabelle. Die 
Tabelle wird im virtuellen Speicher gelöscht. 

Die Schreibvorgänge finden über den symbolischen Namen "ISPTABL" statt. Lesevorgänge werden 
über "ISPTLIB" durchgeführt. In der Praxis zeigen beide Namen auf die gleiche Datei. 

Nachdem DMS-Tabellen als Member von PO-Dateien geführt werden, muss (speziell bei größeren 
Tabellen) die Problematik der Mitgliedsleichen beachtet werden. Um nicht permanente Verarbei-
tungsabbrüche wegen zu kleiner Dateien hinnehmen zu müssen, kann  beim Schreiben zweierlei 
versucht werden: 

Mit REPLCOPY wird versucht, die Tabelle an die Stelle zurück zu schreiben, von der sie gelesen 
wurde. Ist dies aus Platzgründen nicht möglich, wird die Tabelle am Ende angefügt (die Mitgliedslei-
che ist erzeugt). 

Über "PAD(prozentwert)" kann beim ersten Schreiben, und immer dann, wenn sich der Standort der 
Tabelle verändert, ein bestimmter Prozentualwert im Verhältnis zur aktuell benötigten Größe als 
Vorhalt angegeben werden. Findet keine Ortsveränderung der Tabelle statt, wird die Angabe igno-
riert. 

4.14.7. Temporäre Tabellen 

existieren ausschließlich im virtuellen Speicher. bei Ende der Tabellenverarbeitung werden sie nicht 
unter "ISPTABL" auf Platte geschrieben. Wie bei permanenten Tabellen wird bei Ende der Verarbei-
tung die Tabelle aus dem virtuellen Speicher gelöscht. 

Temporäre Tabellen können über die DMS-Befehle "TBOPEN tabname" oder "TBCREATE tab-
name" mit dem Attribut "NOWRITE" erzeugt und durch "TBEND tabname" im virtuellen Speicher 
gelöscht werden. 

4.14.8. Tabellenverarbeitende Befehle 

Über eine Vielzahl von Befehlen kann die Bearbeitung eines Tabellensatzes erfolgen. Für einige 
Befehle ist Bedingung, dass die einzutragenden Feldinhalte in gleichnamigen Variablen im Function-
Pool vorbereitet sind. Die Basisbefehle sind: 

4.14.8.1. TBTOP 

Positioniert den CRP vor den ersten Datensatz. Dies ist immer sinnvoll, wenn ein Suchvorgang in 
einer Tabelle gestartet werden soll und nicht sicher ist, ob der Satzzeiger im Moment auf Tabellen-
anfang steht. 

4.14.8.2. TBBOTTOM 

setzt den Satzzeiger auf den letzten Datensatz und überführt dessen Feldinhalte in Variable in den 
Function-Pool, sofern nicht NOREAD angegeben ist. 

4.14.8.3. TBEXIST 

erlaubt die Überprüfung, ob ein Eintrag in einer Tabelle mit Schlüssel bereits existiert. Ist dies der 
Fall, ist im Anschluss an TBEXIST die Variable, die den Return Code enthält auf 0 gesetzt. 

4.14.8.4. TBDELETE 

führt zum Löschen des Tabellensatzes auf den der CRP zeigt. 

4.14.8.5. TBADD 

erlaubt das Hinzufügen eines Tabellensatzes. Mit der Angabe "ORDER" wird der Satz bei einer 
sortierten Tabelle an der entsprechenden Position eingearbeitet. ORDER sollt immer benutzt wer-
den, da es im Falle einer unsortierten Tabelle keine negativen Auswirkungen hat, ohne ORDER im 
Falle einer sortierten Tabelle die Sortierung kaputt geht. 

4.14.8.6. TBPUT 

führt zur Veränderung des Tabellensatzes, auf den der CRP zeigt.  

Bei Tabellen mit Schlüssel wird die Veränderung nur durchgeführt, wenn die Schlüsselfelder des 
aktuellen Tabellensatzes inhaltlich mit den gleichnamigen Variablen im Function-Pool übereinstim-
men. 

Bei Tabellen ohne Schlüssel wird der Satz geändert, auf den der CRP zeigt. 

Es ist also generell dafür zu sorgen, dass der CRP auf den zu ändernden Tabellensatz positioniert 
wird. 

4.14.8.7. TBMOD 

Vereint die Funktionen aus TBADD und TBPUT.  

Zunächst wird geprüft, ob die gesuchte Zeile in der Schlüsselfeldtabelle enthalten ist. Ist dies der 
Fall, wird der betreffende Tabellensatz aktualisiert. Anderenfalls wird ein neuer Satz in die Tabelle 
eingefügt. 

TBMOD ist bei Tabellen ohne Schlüssel gleichbedeutend mit TBADD. 

4.14.8.8. TBSKIP 

verändert den Standort des CRP nach oben oder unten innerhalb der Tabelle und überstellt die 
Inhalte aller Felder in gleichnamige Variable im Function-Pool. 

4.14.8.9. TBGET 

liefert beispielsweise die Namen so genannter SAVE-Variabler, die nur bei einzelnen Sätzen ange-
hängt wurden. 

4.14.8.10. TBERASE 

löscht eine Tabelle aus ISPTABL, sofern sie nicht innerhalb der gleichen Datei mit dem Attribut 
WRITE in einem TBOPEN-Befehl eröffnet wurde 

4.14.9. Suche in Tabellen 

Um eine Tabelle nach bestimmten Kriterien zu verarbeiten, ohne dabei zu einem sequentiellen Le-
sen in einer Programmschleife gezwungen zu sein, werden zwei Befehle angeboten, die es erlauben 

eine Tabelle so zu betrachten, als wären nur Sätze mit bestimmten Feldinhalten vorhanden. Alle 
Tabellenzeilen, auf die bestimmte Angaben nicht zutreffen werden als nicht vorhanden betrachtet. 
Diese Möglichkeit erleichtert Suchvorgänge innerhalb der Tabelle ganz wesentlich. Die beiden mög-
lichen Verfahren sind: 

Zunächst werden durch "TBVCLEAR" alle Variablen im Function-Pool für die es innerhalb der geöff-
neten Tabelle gleichnamige Felder gibt, gelöscht. 

Anschließend werden die Felder und deren Inhalte, nach denen gesucht werden soll in Form von 
Variablen in den Function-Pool eingetragen. Mit der Instruktion "TBSARG" (TaBleSearchARGu-
ment) wird dem nachfolgenden "TBSCAN" mitgeteilt, dass das Scannen der Tabelle über ein vorbe-
reitetes Suchargument erfolgen muss. 

"TBSCAN" kann nun in einer Verarbeitungsschleife abgesetzt werden. Vorausgesetzt, der CRP wird 
vor Beginn dieser Verarbeitung auf den Tabellenanfang gesetzt, kann die Verarbeitung über den 
Return Code des "TBSCAN" gesteuert werden. 

Das folgende Beispiel zeigt ein REXX-Programm zur Abarbeitung einer Tabelle bei der nach be-
stimmten Sätzen gesucht werden soll: 

/* REXX * TABUPD  

*/ 

address ispexec 

"DISPLAY PANEL(panelname)" 

/* Angenommen: Im Panel wird das Feld "GRUPPE" gefüllt, 

   das mit ARTGRU aus der Tabelle gleichzusetzen ist */ 

"TBOPEN ARTIKEL" 

"TBVCLEAR ARTIKEL" 

ARTGRU=GRUPPE 

"TBSARG ARTIKEL" 

"TBSCAN ARTIKEL" 

DO WHILE RC = 0 

   /* Hier liegt die Verarbeitung */ 

   "TBSCAN ARTIKEL" 

END 

"TBCLOSE ARTIKEL" 

EXIT 

Ergänzend zu diesem standardisierten Verfahren gibt es zwei Besonderheiten, über die die Such-
vorgänge in der Tabelle gesteuert werden können: 

Generic Search: Sollen alle Sätze mit Postleitzahlen, die mit "80" beginnen gesucht werden ist ein 
füllen der entsprechenden Variablen in der Form "PLZ='80*'" möglich. 

Bedingtes Suchargument: Durch "TBSARG tabname NAMECOND(PLZ,GE)" werden alle Sätze 
zur Verarbeitung ausgewählt, bei denen die gesetzte Bedingung erfüllt ist. 

Als Vergleichsoperationen für die bedingte Suche können die üblichen Operanden eingesetzt wer-
den: EQ, NE, LE, LT, GE und GT 

4.14.10. TBSCAN suchargument 

Werden im "TBSCAN" die Suchargumente direkt angegeben, können sowohl "TBVCLEAR" als auch 
"TBSARG" entfallen. Sollten zum Zeitpunkt des Scanvorganges entsprechende Variable im Func-
tion-Pool eingetragen sein, werden sie nicht beachtet. 

Betrachten wir unter diesen Voraussetzungen das gleiche Beispiel 

/* REXX * TABUPD *************************/ 

address ispexec 

"DISPLAY PANEL(panelname)" 

"TBOPEN ARTIKEL" 

ARTGRU=GRUPPE 

DO FOREVER 

 "TBSCAN ARTIKEL ARGLIST(ARTGRU)" , 

 "CONDLIST(EQ)" 

 IF RC > 0 THEN LEAVE 

 /* Hier liegt die Verarbeitung */ 

END 

"TBCLOSE ARTIKEL" 

EXIT 

4.14.11. Positionierung des Satzzeigers (CRP) 

Bei Benutzung diverser Tabellenkommandos sind die Angaben POSITION und ROWID möglich. In 
beiden Fällen wird der Name einer Variablen angegeben, in die der Dialog Manager den Wert des 
CRP hinterlegt. Die Variablen enthalten bei 

▪ ROWID den Wert des CRP vor Ausführung des Befehles 
▪ POSITION den Wert des CRP nach Befehlsausführung 

Beide Operanden sind verfügbar bei den Befehlen 

• TBBOTTOM 

• TBDISPL 

• TBGET 

• TBSCAN 

• TBSKIP 

Wie nach einem "TBSCAN" auf den Tabellensatz rückpositioniert werden kann, auf dem der CRP 
vor "TBSCAN"  stand zeigt das Beispiel  

"TBSCAN tabname ROWID(OLDCRP)" 

: 

"TBSKIP tabname ROW(&OLDCRP)" 

4.14.12. Tabellenanzeige (TBDISPL) 

Panels, die über den Service TBDISPL angezeigt werden, weichen in ihrem Aufbau von normalen 
Datenpanels ab. 

Im )BODY wird der konstante Teil des Bildschirmes beschrieben (Bildschirm-Titelzeile, Kommando-
zeile, SCROLL-Feld und die Spalten-Überschriften) 

Im )MODEL wird das Muster für die Abbildung eines Satzes beschrieben. Dieser Satz kann sich 
über maximal acht Bildschirmzeilen erstrecken. Dieser variable Teil der Maske wird bis zum Bild-
schirmende hin dynamisch erweitert. 

Das 2.Feld im )BODY muss vier Stellen lang sein und enthält den Wert für das Scrolling. Der Feld-
name ist wahlfrei. 

Soll eine Tabelle nur angezeigt werden, müssen alle Felder mit dem Attribut OUTPUT belegt wer-
den. 

Sollen nicht alle Sätze über den TBDISPL angezeigt werden, können durch den TBSARG die Such-
argumente festgelegt werden. Im Panel des TBDISPL kann durch eine Angabe ROWS(SCANN) in 
der )MODEL-Zeile erreicht werden, dass nur Sätze gezeigt werden, die dem Suchargument entspre-
chen. 

Werden gültige ISPF-Kommandos oder Scroll-Werte eingegeben, bleibt die Steuerung beim 
TBDISPL, der die entsprechenden Aktionen durchführt. 

Mit der END-Funktion, oder wenn ein nicht gültiges ISPF-Kommando eingegeben wird, geht die 
Steuerung an die aufrufende Funktion zurück. 

4.14.13. Tabellen-Änderungen 

Nachdem zur Anzeige einer Tabelle nur eine einzige Musterzeile im )MODEL-Teil definiert wurde, 
werden alle Tabellensätze über die Feldnamen dieser einen Musterzeile verarbeitet. Diese Situation 
würde normalerweise bei Full-Screen-Updates ziemliche Probleme aufwerfen. Hier setzt ein weiterer 
Komfort des Dialog Managers ein: 

In der DMS-Systemvariablen &ZTDSELS wird die Anzahl der im Panel gezeigten, geänderten Zeilen 
abgelegt, wenn die ENTER-Taste gedrückt wurde. 

Nach <ENTER> steht der CRP auf dem ersten modifizierten Satz. Die Feldinhalte stehen in gleich-
namigen Variablen im Function-Pool. 

Durch nachfolgende TBDISPL ohne Panelangabe wird der CRP auf den nächsten modifizierten Satz 
gesetzt und &ZTDSELS um 1 vermindert. In dieser Form wirkt der TBDISPL wie ein Scanvorgang. 

4.14.14. Generelles Tabellenhandling 

In der Praxis wird häufig ein Verfahren angewandt, das im ersten Schritt die gewünschten Sätze 
anzeigt (alle Felder OUTPUT). Vergleichbar zu den EDIT-LineCommands wird vor der Tabellenzeile 
ein Feld gesetzt, über welches die Verarbeitung gesteuert wird (z.B.: A=Ändern, L=Löschen etc.). 

 Zu beachten ist, dass es sich bei diesem Feld meist nicht um ein Tabellenfeld handelt und es somit 
durch den Befehl TBVCLEAR nicht berührt wird. 

Auch in dieser Form wird über die Variable &ZTDSELS erkannt, für wie viele Tabellenzeilen das 
Verarbeitungskennzeichen gesetzt wurde. Die Betreffenden Zeilen können analog zur vorangegan-
genen Beschreibung in einer Schleife abgearbeitet werden. 

4.14.14.1. Systemvariable bei Tabellenanzeige 

Um diverse brauchbare Informationen innerhalb der Tabellenanzeige zur Verfügung zu haben, kön-
nen einige Systemvariable des Dialog Managers genutzt werden: 

Systemrvariable Bedeutung 

&ZTDSELS Enthält die Anzahl der modifizierten Tabellensätze 

&ZTDTOP Übergibt die Nummer des obersten am Bildschirm sichtbaren Tabellensatzes 

&ZTDROWS Liefert die Anzahl der Tabellenzeilen die am Bildschirm abgebildet werden 
können (wie im )MODEL-Teil beschrieben) 

&ZTDMARK Kann eine andere Schlusszeile als die standardgemäße "***BOTTOM OF 
DATA***" erzeugen. Diese Variable wird im )INIT-Teil gefüllt. Häufig wird der 

zuzuordnende Wert aber bereits in REXX in einer Variablen hinterlegt, da hier 
wesentlich komfortablere Funktionen zur Stringbehandlung zur Verfügung 
stehen. 

Diese Variablen stellen lediglich einen Auszug dar. Weitere Systemvariable 
des Dialog Managers können im Teil SYNTAX nachgelesen werden. 

 

Diese Variablen stellen lediglich einen Auszug dar. Weitere Systemvariable des Dialog Managers 
können im Teil SYNTAX nachgelesen werden. 

4.14.15. Panel für Tabellenanzeige 

)ATTR DEFAULT(%+_) 

   µ TYPE(TEXT)   HILITE(USCORE) 

   $ TYPE(OUTPUT) INTENS(LOW) CAPS(OFF) 

)BODY EXPAND(//) 

+Anzeige der Kundentabelle /-/ 

+COMMAND ==>_ZCMD/ /+SCROLL ==>_SAMT 

+ 

µKdNr µName       µVorname          µAnschrift 

)MODEL 

$KDNR $KDFNAME    $KDVNAME          $KDADR1 

+                                   $KDADR2 

+                                   $KDADR3 

+                                   --------- 

)INIT 

   &ZTDMARK = &ENDMARK 

   IF (&SAMT = &Z) 

      &SAMT = HALF 

)PROC 

   VPUT (SAMT) PROFILE 

)END 

4.14.16. Zugriffsschutz 

Um das leidige Problem konkurrierender Updates zu erledigen, sperrt Dialog Manager die betref-
fende Tabelle (Member einer PO-Datei) mittels ENQ automatisch in der Weise: 

• Tabellen für die ein Update vorgesehen ist (WRITE-Option beim Eröffnen angegeben) wer-
den ab dem Zeitpunkt des Öffnens bis zum Schließen gesperrt. 

• Temporäre Tabellen (mit NOWRITE eröffnet) werden während der Einlesephase in den vir-
tuellen Speicher gesperrt. 

Alles in allem ist dieser Zugriffsschutz seitens des Dialog Managers aber als sehr dürftig anzusehen, 
da unter ISPTLIB häufig mehrere Dateien verkettet sind, Dialog Manager sich mit seinen Schutzme-
chanismen aber ausschließlich um die Erste kümmert. 

Wenn ein Dialog sehr umfangreich mit Tabellen arbeitet und sehr viele Benutzer damit umgehen 
müssen, empfiehlt es sich, die Datenhaltung gegebenenfalls ins DB2 zu verlegen und innerhalb des 
Dialog Managers nur noch temporäre Tabellen zur Datenanzeige zu benutzen. 

4.15. Skeletons 

Skeletons werden vom Dialog Manager im File-Tailoring-Service verarbeitet. Es handelt sich hierbei 
um Musterdateien, die eingelesen und zeilenweise interpretiert werden. In diesen Musterdateien 
(letztlich immer Member von PO-Dateien) können sowohl Steueranweisungen als auch Daten hin-
terlegt sein. Diese Skelette können vom File Tailoring Service beliebig zusammengebunden, und in 
eine Datei ausgegeben werden. Die Übergabe der Daten in die Ausgabedatei kann  im Verhältnis 
1:1 stattfinden, oder durch entsprechende Steueranweisungen individuell geregelt werden. 

Einige Merkmale, die die Stärke des Services aufzeigen sind: 

▪ Einsatz von Variablen in Skelettdateien 

▪ Durch eingebettete Steueranweisungen in einer Skelettdatei weitere Skelettdateien zu im-
plementieren 

▪ Abhängig von Bedingungen auch einzelne Datensätze in eine Skelettdatei einzubauen 

▪ Tabellen des Dialog Managers zu verarbeiten 

Skelette müssen Member einer PO-Datei sein, die unter dem Namen ISPSLIB verkettet ist. Die Aus-
gabe des File Tailoring Service wird in einer Datei hinterlegt, die unter dem Namen ISPFILE zuge-
wiesen sein muss. 

4.15.1. File Tailoring Dienste 

4.15.1.1. FTOPEN [TEMP] 

ruft den File Tailoring Service auf. Wird TEMP angegeben, wird erreicht, dass die Ausgabe nur in 
Form einer temporären Datei erzeugt wird, deren Name über die Systemvariable &ZTEMPF des 
Dialog Managers in Erfahrung gebracht werden kann. Sollte diese Ausgabe JCL-Statements enthal-
ten, kann die Ausführung durch den TSO-Befehl Submit erfolgen. 

4.15.1.2. FTINCL member 

fügt das angegebene Member aus einer unter ISPSLIB zugewiesenen PO-Datei in den Verarbei-
tungsablauf ein. Generell können in dieser Form mehrere Member aus der Skelett-Bibliothek nach-
einander eingebunden werden, wobei jedes dieser Member wiederum intern weitere Member aufru-
fen kann. 

4.15.1.3. FTCLOSE 

beendet den Service und führt zur Übergabe der Daten in die vorgesehene Ausgabedatei. 

4.15.1.4. File Tailoring Steuerbefehle 

In Skelettdateien können Steueranweisungen enthalten sein. Diese Steueranweisungen müssen 
zwingend in Spalte 1 kodiert werden und mit einer ) beginnen. Die Schlüsselworte sind zwingend 
groß zu schreiben. 

Steuerbefehle Bedeutung 

)DEFAULT Änderung von Steuerzeichen 

)TB oder )TBA Setzen des Tabulators 

)IM Einbetten weiterer Skeletons 

)IF THEN Abfrage einer Bedingung (JA-Zweig) 

)ELSE NEIN-Zweig einer IF-Abfrage 

)SEL Abfrage einer Bedingung (ohne NEIN-Zweig) 

)ENDSEL Abschluss einer )SEL-Bedingung 

)DO Beginn einer Befehlsfolge (meist in Verbindung mit )IF/)ELSE 

)ENDDO Abschluss einer Befehlsfolge 

)DOT Beginn einer Tabellenverarbeitungsschleife 

)ENDDOT Ende einer Tabellenverarbeitungsschleife 

)SET Wertzuweisung 

)BLANK [zahl] Ausgabe von Leerzeilen 

)REXX Beginn REXX-Code 

)ENDREXX Ende des REXX-Codes 

)CM Kommentare, die nicht verarbeitet werden 

 

4.15.1.4.1. )DEFAULT )&?!<|> 

Erfüllt grundsätzlich den gleichen Zweck, wie die entsprechende Kodierung im )ATTR-Teil eines 
Panels. Es werden einige Sonderzeichen festgelegt, die im weiteren Ablauf bestimmte Aussagen 
treffen. Die obenstehenden Zeichen entsprechen dem Standard. Soll davon abgewichen werden, 
sind immer alle sieben Zeichen zu definieren. Die einzelnen Stellen haben folgende Bedeutung: 

Byteposition Bedeutung 

x1 Kennzeichnet eine FT-Steueranweisung 

x2 Benennt eine Variable 

x3 Markiert eine Fortsetzung 

x4 Tabulatorzeichen 

x5 Beginn einer ODER-Übertragung 

x6 Trennzeichen der ODER-Übertragung 

x7 Ende der ODER-Übertragung 

Um beispielsweise für die Trennung der ODER-Übertragung abweichend vom Standard das Dollar-
zeichen zu nutzen ist die Angabe DEFAULT )&?!<$> nötig. 

Die ODER-Übertragung kann eingesetzt werden, wenn eine Übertragung einer Variablen abhängig 
von einem eventuellen Inhalt durchgeführt werden soll. 

Angenommen, es existieren die Variablen TEL1 und TEL2. Wenn TEL1 nur übertragen werden soll, 
sofern es gefüllt ist, anderenfalls die Übertragung von TEL2 zu erfolgen hat, sieht die Kodierung 
folgendermaßen aus: 

<&TEL1$&TEL2> 

4.15.1.4.2. )DOT 

)DOT tabname  

 :  

)ENDDOT 

Führt Tabellenverarbeitung durch (DO Table). Die zwischen den )DOT und )ENDDOT liegenden 
Instruktionen werden je Tabellenzeile ausgeführt. Die Tabelle muss zum Zeitpunkt der Verarbeitung 
durch File Tailoring nicht geöffnet sein. 

4.15.1.4.3. )SEL 

)SEL bedingung  

 :  

)ENDSEL 

Ist vergleichbar mit einer IF-Abfrage. Die Aktionen zwischen )SEL und )ENDSEL werden durchge-
führt, wenn die Bedingung im )SEL-Statement zutrifft, anderenfalls werden sie übersprungen.  

Darüber hinaus können bis zu acht Ausdrücke durch |(oder) bzw. &&(und) miteinander verbunden 
werden. 

4.15.1.4.4. )IF 

)IF bedingung THEN )DO 

 : 

)ENDDO 

)ELSE )DO 

 : 

)ENDDO 

Die IF-THEN-ELSE-Konstruktion ist eine Neuerung seit z/OS 1.8. Es gelten die gleichen Möglich-
keiten wie im )SEL-Befehl. Zusätzlich kann ein NEIN-Zweig kodiert werden. 

4.15.1.4.5. )IM membername 

erlaubt, aus einem Skelett heraus ein anderes aufzurufen. Im Gegensatz zu aufeinander folgenden 
FTINCL-Anweisungen, die die angegebenen Skeletons sequentiell aneinanderreihen, kann die An-
weisung )IM ein Skelett an beliebiger Stelle eingebunden werden. 

Die maximale Tiefe von Verschachtelungen in dieser Form ist auf drei Ebenen begrenzt. 

4.15.1.4.6. )SET|SETF variable=wert 

erlaubt Variable innerhalb von Skelettdateien zu verändern. Als mögliche arithmetische Operatoren 
sind nur plus(+) und minus(-) erlaubt. Vergleichbare Möglichkeiten wie in der SET-Anweisung im 
Panel oder der Wertzuweisung in REXX existieren nicht. 

Ein gängiger Anwendungsbereich ist aber das Zählen von Ausgabezeilen, um entsprechende Sei-
tenvorschübe bei einer Druckausgabe zu regeln. 

Die inhaltliche Länge numerischer Variabler ist auf 8 Byte begrenzt. 

4.15.1.4.7. )TAB 

)TB  zahl[A] [zahl[A] .. ]  

)TBA zahl [zahl .. ] 

setzt Tabulatoren für die Ausgabe. So kann erreicht werden, dass Variable ungleicher Länge spal-
tengerade ausgegeben werden können. 

Es können bis zu 16 Tabulator-Haltepunkte definiert werden. Der Tabulatorsprung wird standardge-
mäß durch das Ausrufezeichen(!) ausgeführt. 

Bei sehr unterschiedlich langen Variableninhalten besteht die Möglichkeit, nach Variablensubstitu-
tion einen oder mehrere Tabulatorhaltepunkte durch die Länge des Textinhaltes zu überdecken und 
so durch den Tabulatorsprung die zulässige Satzlänge zu überschreiten. In diesem Fall wird die 
Verarbeitung mit dem Hinweis "OUTPUT BUFFER OVERFLOW" abgebrochen. 

Um dieses Problem zu umgehen, oder aber zumindest an der Wurzel zu erkennen, ist empfehlens-
wert, den Ausgabesatz feldweise aufzubauen und nach jedem Feld einen Testlauf durchzuführen. 

Alternative Tabulatorsprünge (TBA.. oder TB xx xxA xx) erzwingen eine Positionierung auf die an-
gegebene Stelle. Im Bedarfsfall wird das vorgelagerte Feld beschnitten. 

4.15.1.4.8. )BLANK [zahl] 

Soll in der Ausgabe eine Leerzeile erzeugt werden, ist dies nur über die )BLANK-Anweisung möglich. 
Wird ein Ausgabesatz über Variable aufbereitet und diese Variablen sind leer, wird an dieser Stelle 
keine Leerzeile erzeugt, sondern die entsprechende Zeile  wird unterdrückt. 

Der )BLANK-Anweisung kann in Form einer integeren Zahl die Anzahl der einzustreuenden Leer-
zeilen mitgegeben werden. Dieser Wert kann auch über eine Variable geliefert werden. 

4.15.1.4.9. )REXX 

)REXX var1,var2,varn 

 : 

)ENDREXX 

Erlaubt die Einbettung von REXX-Funktionalität im Skeleton. 

4.15.1.4.10. )CM 

kennzeichnet die betreffende Zeile als Kommentar. Diese Zeilen werden bei Verarbeitung nicht wei-
ter berücksichtigt, können aber eine Hilfe für Veränderungen in einer Skelettdatei darstellen. 

Wie mit /*-*/ in CLIST oder REXX können Kommentare als Online-Dokumentation benutzt werden 
und tragen letztlich dazu bei, eine gewisse Änderungsfreundlichkeit zu gewährleisten. 

4.15.1.5. FT-Beispiel 

Das nachfolgende Beispiel zeigt das Wirken der Steuerbefehle im Verbund. Über implementierte 
JCL werden die Sätze einer Tabelle zur Druckausgabe gebracht: 

: 

"FTOPEN" 

"FTINCL SKEL01" 

"FTCLOSE NAME(member)" 

: 

 

Inhalt von SKEL01: 

)IM SKEL02 

)TB 20 40 46 

)SET SEITE = 0 

)SET ZEILEN = 99 

)DOT ADRTAB1 

)SEL &ZEILEN > 65 

)SET SEITE = &SEITE + 1 

)SET ZEILEN = 4 

Ausgabe Adressenliste vom &ZDATE - Seite: &SEITE 

)BLANK 3 

)ENDSEL 

&NAME ! &VNAME ! &PLZ ! &ORT &STRASSE 

)SET ZEILEN = &ZEILEN + 1 

)ENDDOT 

 

Inhalt von SKEL02: 

//&ZUSER.A JOB (&ACCTNR),'&ZUSER',NOTIFY=&ZUSER 

/*JOBPARM LINECT=65 

//TABPRINT EXEC PGM=IEBGENER 

//SYSIN    DD DUMMY 

//SYSPRINT DD SYSOUT=* 

//SYSUT2  DD SYSOUT=* 

//SYSUT1  DD * 

4.15.2. Bibliothekszuweisungen 

Häufig wird in einem Dialog ein Umfeld benötigt, das über die Standardzuordnungen nicht erreichbar 
ist. Die erforderlichen Zuordnungen mit ALLOCATE durchzuführen macht in der Praxis wenig Sinn, 
da dabei die bereits bestehenden Angaben in der Zuordnungstabelle überschrieben werden. 

Hier setzt ein Service ein, der erlaubt, die Zuordnung von Dateien unter bestimmten DD-Namen 
dynamisch durchzuführen. Dabei werden die aktuellen Zuordnungen unter dem entsprechenden 
DD-Namen nicht überschrieben, sondern die angegebenen Dateien werden in der Zuordnungsta-
belle unter dem angegebenen DD-Namen vorne eingefügt. 

Das nachfolgende Beispiel zeigt die dynamische Veränderung der Zuweisung einer Datei unter dem 
Namen ISPPLIB. Dialog Manager wird im Anschluss daran beim Aufruf von Panels diese Datei als 
erste nach einem gleichnamigen Member durchsuchen. 

"LIBDEF ISPPLIB DATASET ID('SP.EDU.PANELS')" 

4.15.2.1. Adressierung 

Um die Dienste des Dialog Managers aus einer Funktion heraus nutzen zu können, müssen die 
Aufrufe an die richtige Systemumgebung durchgeführt werden. Diese Umgebung heißt ISPEXEC. 
Die Adressierung in REXX wird in nachfolgendem Beispiel eines Panelaufrufes verdeutlicht: 

ADDRESS ISPEXEC "DISPLAY PANEL(pname)" 

Grundsätzlich ist diese Form der Inanspruchnahme von Dialog Manager Diensten mit viel Kodierauf-
wand verbunden, da die standardgemäß eingestellte Systemumgebung das TSO ist. Man sollte sich 
daher grundsätzlich überlegen, ob es nicht sinnvoll ist, diese Umgebung auf ISPEXEC umzuschalten 
und im Bedarfsfall TSO-Kommandos (sofern sie überhaupt benutzt werden), über die ADDRESS-
Instruktion an das TSO zu übergeben. 

/* REXX * PROG001 *************/ 

ARG TEST 

IF TEST = 'T' THEN TRACE '?R' 

ADDRESS ISPEXEC 

"LIBDEF ISPPLIB DATASET ID('dateiname')" 

: 

"DISPLAY PANEL(pname)" 

IF RC = 0 THEN DO 

 ADDRESS TSO "DEL '"datei"'" 

END 

ELSE ... 

4.16. ISPF-Demos 

Die nächsten Seiten zeigen an diversen Beispielen, wie die Aufrufe an den Dialog Manager aus 
einem REXX-Programm heraus aussehen können, je nach Bedarf und Situation: 

4.16.1. Panelaufruf 

"DISPLAY PANEL(pname)" 

IF RC = 0 THEN DO 

 /* Hier ist gewährleistet, dass das Panel durch */ 

 /* die E-Taste verlassen wurde und   */ 

 /* alle Prüfungen im )PROC-Teil erfolgreich  */ 

 /* durchgeführt wurden                       */ 

END 

ELSE DO 

 /* Kann mehrere Gründe haben. Die unterschied- */ 

 /* lichen Möglichkeiten können beim DISPLAY- */ 

 /* Service im Syntax-Teil nachgelesen werden. */ 

END 

4.16.1.1. Plausibilitätsprüfung in der Function 

Können bestimmte Prüfungen nicht im Panel vollzogen werden, kann die Plausibilitätskontrolle 
REXX erforderlich werden. Dieses Verfahren ist nach Möglichkeit zu umgehen, lässt sich aber nicht 
immer vermeiden. In diesem Fall muss die Panelanzeige in Schleife stattfinden, deren Ende über 
REXX gesteuert wird. Im Beispiel unterstellen wir, dass im Panel ein Feld DATUM mit einem Datum 
gefüllt wurde. Eine Funktion DATCHK soll nun prüfen, ob der entsprechende Inhalt der Variablen 
einem gültigen Datum entspricht. Ist dies der Fall, kann das Programm fortgesetzt werden, ande-
renfalls wird das Panel wieder angezeigt und eine entsprechende Fehlermeldung ausgegeben: 

DO FOREVER 

 "DISPLAY PANEL(pname)" 

 IF DATCHK(VDAT1) = 'OK' THEN LEAVE 

 "SETMSG MSG(ERR001A)" 

END 

4.16.1.2. REXX-Logik im Panel als Subroutine 

Vorausgesetzt im )BODY wird ein Feld DATUM 

)PROC 

if (.RESP = END) exit 

*REXX(tt,mm,jj,datum) 

   parse var datum 1 tt 3 p1 4 mm 6 p2 7 jjjj 11 . 

   select 

      when  p1 !! p2 <> ".." then do  

         langmsg = "Keine Punkte an Stelle 3 und/oder 6" 

         zrxmsg  = "TRY000E" 

         zrxrc   = 8 

      end 

      when ^datatype(tt !! mm !! jjjj,"W") then do 

         zrxmsg  = "TRY000D" 

         zrxrc   = 8 

      end 

      when jjjj !! mm !! tt < 15281015 then do 

         zrxmsg  = "TRY000C" 

         zrxrc   = 8 

      end 

      when mm < 1 ! mm > 12 then do 

         zrxmsg  = "TRY000B" 

         zrxrc   = 8 

      end 

      otherwise do 

         tim = "31" , 

                28+(jjjj//4=0)-(jjjj//100=0)+(jjjj//400=0) , 

               "31 30 31 30 31 31 30 31 30 31" 

         if tt > word(tim,mm) then do 

            zrxmsg  = "TRY000A" 

            zrxrc   = 8 

         end 

         else zrxrc = 0 

      end 

   end 

   return 

*ENDREXX 

)END 

 

Zwischen *REXX und *ENDREXX kann beliebiges REXX-Codeing liegen. Der Ablauf kann mittels 
TRACE auch verfolgt werden. 

4.16.1.3. REXX-Logik im Panel als externe Routine 

Vorausgesetzt im )BODY wird ein Feld DATUM. 

)PROC 

if (.RESP = END) 

   exit 

*REXX(*,(RXDATCHK)) 

)END 

Die Prüfung kann auch als externe Routine irgendwo unter SYSPROC/SYSEXEC liegen. Dies hat 
den Vorteil, dass sie in beliebigen Panels oder anderen REXX-Programmen benutzt werden kann. 

4.16.1.4. ISPF aktiv? 

Sollte ein Programmaufruf unter Umständen nicht unterhalb ISPF/PDF erfolgen, muss Dialog Ma-
nager erst aktiviert werden, bevor die Dienste in Anspruch genommen werden können. Diese Situ-
ation lässt sich in REXX über die Systemvariable SYSISPF in Erfahrung bringen: 

IF SYSVAR(SYSISPF) <> "AKTIV" THEN DO 

 "ISPSTART CMD(startpgm)" 

END 

ELSE DO 

 ADDRESS ISPEXEC "SELECT CMD(startpgm)" 

END 

4.16.1.5. Öffnen Tabelle 

Tritt beim Table OPEN ein Problem auf, kann dies durch RC im Anschluss an den Befehl erkannt 
werden. Das nachfolgende Beispiel zeigt, wie bei Erkennen der Situation "Tabelle nicht vorhanden" 
die Neuanlage der Tabelle veranlasst werden kann: 

"TBOPEN tabname WRITE" 

SELECT 

 WHEN RC = 0 THEN NOP /* OPEN war erfolgreich   */ 

 WHEN RC = 8 THEN DO 

  /* Tabelle wird anderweitig belegt     */ 

 END 

 WHEN RC = 12 THEN DO  

  /* Tabelle existiert nicht, Neuanlage  */ 

  "TBCREATE tabname KEYS(..) NAMES(..)" 

 END 

 WHEN RC = 16 THEN DO 

  /*Tabellen-Bibliothek nicht zugewiesen */ 

 END 

 OTHERWISE DO 

  /* RC 20, Schwerer Fehler */ 

 END 

END  

4.16.1.6. Tabellensatz einfügen 

In einer Tabelle soll ein neuer Datensatz eingefügt werden, sofern das Erfassungspanel mit der 
ENTER-Taste verlassen wird. Die entsprechenden Variablen werden über ein Panel versorgt. Die 
Tabelle hat Schlüsselfelder. Es soll ein Hinweis erfolgen, ob das Einfügen des Satzes möglich, oder 
nicht möglich war.  

DO FOREVER 

 "DISPLAY PANEL(pname)" 

 IF RC > 0 THEN LEAVE 

 "TBADD tabname" 

 IF RC = 0 THEN DO 

  /* Einfügen des Tabellensatzes erfolgreich */ 

  "SETMSG MSG(MSG001)" 

 END 

 ELSE DO 

  /* Einfügen war nicht möglich, Satz mit   */ 

  /* identischen Schlüsselfeldeintragungen  */ 

  /* in Tabelle bereits vorhanden           */ 

  "SETMSG MSG(MSG002)" 

 END 

END 

4.16.1.7. Satz-Selektion in der Tabelle 

Tabellenanzeige mit Line-Command-Feld. Mögliche Optionen: E(infügen), Z(eigen) und L(öschen). 
Die markierten Sätze werden über &ZTDSELS verarbeitet. Abhängig von der Funktion werden die 
erforderlichen Aktionen in einer Subroutine abgehandelt. 

"TBDISPL tabname PANEL(pname)" 

DO WHILE RC <= 4 

 DO WHILE ZTDSELS > 0 

  "CONTROL DISPLAY SAVE" 

  SELECT 

   WHEN OPT = 'E' THEN CALL INS 

   WHEN OPT = 'A' THEN CALL UPD 

   WHEN OPT = 'L' THEN CALL DEL 

   OTHERWISE NOP 

  END 

  "CONTROL DISPLAY RESTORE" 

  IF ZTDSELS > 1 THEN "TBDISPL tabname" 

  ELSE ZTDSELS = 0 

 END 

 "TBDISPL tabname PANEL(pname)" 

END 

"TBCLOSE tabname" 

EXIT 

INS: /* Subroutine INS * Einfügen eines Tabellensatzes */ 

 "TBVCLEAR tabname" 

 "DISPLAY PANEL(pname)" 

 IF RC = 0 THEN DO 

  "TBADD tabname ORDER" 

  IF RC = 0 THEN "SETMSG MSG(msgnr)" 

  ELSE "SETMSG MSG(msgnr)" 

 END 

RETURN 

UPD: /* Subroutine UPD * Ändern eines Tabellensatzes */ 

 "DISPLAY PANEL(pname)" 

 IF RC = 0 THEN DO 

  "TBPUT tabname ORDER" /* mit KEYS TBMOD */ 

  "SETMSG MSG(msgnr)" 

 END 

RETURN 

DEL: /* Subroutine DEL * Löschen eines Tabellensatzes */ 

 "DISPLAY PANEL(pname)" 

 IF RC = 0 THEN "TBDELETE tabname" 

RETURN 

4.16.1.8. Verstreute Meldungen 

Eine Subroutine kann viel Zeit in Anspruch nehmen (wenn beispielsweise große Dateien verarbeitet 
werden). In diesem Beispiel wird eine Datei ausgelesen und als Tabelle abgelegt. Der Anwender 
soll mit Hilfe eines POP-UP's informiert werden, bei welchem Anfangsbuchstaben die Verarbeitung 
momentan ist: 

/* Subroutine Datei in Tabelle */ 

DAT_TAB: 

 zwinttl = "Import" 

 last = "" 

 "EXECIO * DISKR ddname (FINIS" 

 do queued() 

  parse pull satz 

  if left(satz,1)<>last then do 

   last = left(satz,1) 

   text = "Verarbeite Buchstabe" last 

   "ADDPOP ROW(10) COLUMN(13)" 

   "CONTROL DISPLAY LOCK" 

   "DISPLAY PANEL(pname)" 

   "REMPOP" 

  end 

  /* Hier wird der Tabellensatz vorbereitet */ 

  "TBADD tabname ORDER" 

 end  

RETURN 

Der Anwender hat vom Auswahlmenu "Datei-Import nach Tabelle" gewählt und sieht im POP-UP, 
wo die Verarbeitung steht.  

4.16.2. Kommandotabellen 

Erfolgt in der Kommandozeile eines ISPF-Panels eine Eingabe, kann die standardgemäße Steue-
rung des Panels verlassen werden, um das entsprechende Kommando auszuführen. Welche Aktio-
nen auf eine bestimmte Eingabe hin durchzuführen sind, wird über verschiedene Kommandotabellen 
gesteuert, die in folgender Reihenfolge durchsucht werden: 

• eigene, applikationsabhängige Kommandotabelle 

• Systembezogene Kommandotabelle 

• in Verbindung mit EDIT auch SYSPROC/SYSEXEC für das Auffinden von EDIT-Makros 

Durch entsprechende Angaben lassen sich auf diesem Weg Systemkommandos unwirksam ma-
chen, ALIAS-Namen vergeben, oder eigene Befehle hinterlegen. 

Wird beim Aufruf eines Dialoges keine Applikationskennung mitgegeben, benutzt ISPF/PDF stan-
dardgemäß die Tabelle ISRCMDS. Im Bedarfsfall kann über das Schlüsselwort NEWAPPL(abc) bei 
ISPSTART oder im SELECT-Service die Verbindung zur jeweiligen Kommandotabelle "abcCMDS" 
hergestellt werden. Die Kommandotabelle ist als Member unter ISPTLIB abzulegen. 

Die Pflege der Kommandotabelle kann menügesteuert über die Option 3.9 des ISPF/PDF oder über 
die Tabellenservices des Dialog Managers erfolgen. 

4.16.2.1. Aufbau der Kommandotabelle 

In einem Satz der Tabelle sind alle Eintragungen hinterlegt, die die Aktionen für ein bestimmtes 
Kommando enthalten. Die Felder eines Tabellensatzes sind 

Feldname Inhalt 

ZCTVERB Befehlsname 

ZCTTRUNC Abkürzungsmöglichkeiten für den Befehl 

ZCTACT Beschreibt die Aktion, welche auszuführen ist 

ZCTDESC Beschreibung des Befehles (optional) 

 
Bei Veränderungen der Kommandotabelle darf die entsprechende Applikation nicht aktiv sein. 

Hütet euch in jedem Fall davor, jede Form von ISPF-Tabelle native im Editor durchzuführen. Diverse 
Informationen liegen HEX vor (z.B.: die aktuelle inhaltliche Länge von Feldern). Die Tabelle ist 
schneller kaputtgeschrieben als repariert. 

4.16.2.1.1. ZCTVERB (CommandTableVERB) 

Der  Befehl kann zwei bis maximal acht Zeichen lang sein. Stelle 1 des Befehles muss ein Buchstabe 
sein.  

Eingetragen werden können die ISPF-Kommandonamen oder eigene Befehle.  

Werden ISPF-Kommandonamen eingetragen, werden die entsprechenden ISPF-Kommandos er-
setzt (das heißt: An Stelle geläufiger ISPF-Kommandos können auch eigene Funktionen aufgerufen 
werden) 

4.16.2.1.2. ZCTTRUNC (Command Table TRUNCation) 

Hier können die möglichen Abkürzungen zum jeweiligen Befehl geregelt werden. Dabei gilt: 

• Der vollständige Befehlsname muss eingegeben werden. Abkürzungen sind nicht erlaubt 

• >=2 Es sind mindestens so viele Zeichen beginnend mit dem ersten einzugeben, als durch 
die Zahl definiert sind.  

• Einstellige Abkürzungen sind nicht zulässig. 

4.16.2.1.3. ZCTACT (Command Table ACTion) 

beschreibt die Aktion, die durchgeführt wird, wenn der entsprechende Befehl in der Kommandozeile 
eines Panels eingetragen wird. 

Als Aktionen sind erlaubt: 

▪ SELECT-Service 

▪ ALIAS 

▪ PASSTHRU 

▪ SETVERB 

▪ NOP 

4.16.2.1.4. SELECT 

SELECT PANEL(pname) 

SELECT CMD(%proc-x) 

SELECT PGM(ISPTUTOR) PARM(ISR0000) 

4.16.2.1.5. ALIAS 

Einem Befehlswort kann ein alternativer Name zugewiesen werden. 

4.16.2.1.6. PASSTHRU 

Das Kommando wird an die aufrufende Funktion weitergeleitet. 

4.16.2.1.7. SETVERB 

Der eingegebene Befehl wird in der ISPF-Systemvariablen ZVERB ohne Parameter abgelegt. Diese 
liegen in ZPARM 

4.16.2.1.8. NOP 

Das Kommando wird deaktiviert. Wird es in der Kommandozeile eingegeben, erscheint im Bereich 
der Short Message der Hinweis "xxxxxxxx is not active" 

4.16.2.1.9. ZCTDESC 

Erlaubt einen Kommentar zum Tabelleneintrag. ZCTDESC ist kein Pflichtfeld. 

4.16.2.2. Beispiel Kommandotabelle 

Unterstellen wir folgende Einträge in einer Kommandotabelle: 

ZCTVERB ZCTTRUNC ZCTACT ZCTDESC 

SELECT 4 SELECT PANEL(&ZPARM) Auswahl rufen 

BACK 2 ALIAS RETURN Anderer Name 

UP 0 PASSTHRU Weiterreichen 

DOWN 0 PASSTHRU Weiterreichen 

TSO 0 NOP Deaktivieren 

4.16.2.2.1. SELECT  

kann verkürzt bis auf 4 Byte eingegeben werden. Als Parameter muss der Name des Panels einge-
geben werden, das über den entsprechenden SELECT-Service aufgerufen werden soll. 

COMMAND ===> SELE ISR@PRIM 

4.16.2.2.2. BACK 

über die Eingabe BACK(BAC oder BA) erfolgt die gleiche Reaktion wie bei RETURN 

4.16.2.2.3. UP/DOWN 

werden nicht zum Dialog Manager durchgereicht, sondern an die ausführende Funktion (zum Bei-
spiel EDIT) übergeben. 

4.16.2.2.4. TSO  

Das Kommando TSO ist im aktuellen Dialog nicht verfügbar. Wird es abgesetzt, erscheint der Hin-
weis ...not active 

4.16.2.3. Kommando-Auswertung 

Die Auswertung von Eingaben in der Kommandozeile einer Bildschirmmaske verläuft immer nach 
dem Schema: 

▪ Durchsuchen der applikationsspezifischen Kommandotabelle. Sofern keine vorhanden ist, 
oder der Befehl dort nicht gefunden werden kann 

▪ Durchsuchen der User-Commandtable. Sofern keine vorhanden ist, oder der Befehl dort 
nicht gefunden wird 

▪ Durchsuchen der Site-Commandtable (firmenspezifisch). Sofern keine vorhanden ist oder 
der Befehl dort nicht gefunden wird 

▪ Durchsuchen der ISPF-Kommandotabelle. Wird der Befehl auch hier nicht gefunden, 
▪ Wird er unverändert an die aufrufende Funktion (REXX, etc.) übergeben. 

Ist ein Befehl in einer der Tabellen vorhanden, kann er trotzdem direkt an die aufrufende Funktion 
übergeben werden, indem der Anwender vor den Befehl das Zeichen GRÖSSER(>) stellt. 

Ob und wie viele Site- oder User-Tables vorhanden sein können, wird in der ISPF-Config-Table 
festgelegt. Wenn der für das ISPF zuständige Kollege etwas "störrisch" ist, könnte ein Benutzer auch 
seine eigene ISPF-Konfiguration festlegen. Wie das von Statten geht, ist im Netz einfach nachzule-
sen. Sucht nach ISPCCONF. 

4.16.3. Steuervariablen 

Steuervariable werden genutzt, um die Anzeige eines Panels oder einer Nachricht abhängig von 
bestimmten Situationen zu steuern. Die Steuervariablen können innerhalb eines Panels nur in den 
Bereichen )INIT, )REINIT und )PROC kodiert werden. 

Die Steuervariablen werden zunächst auf BLANK gesetzt, sobald der Display Service die Steuerung 
des Ablaufes erhält.  

Enthalten die Steuervariablen .MSG, .CURSOR und .CSRPOS nach der Initialisierungsphase des 
Panels keinen Wert, werden die entsprechenden Werte aus der aufrufenden Funktion übergeben, 
sofern hier überhaupt  Angaben erfolgten.  

Wird eine dieser Variablen in der aufrufenden Funktion gefüllt und auch im )INIT-Teil des Panels 
definiert, wird nur die Zuweisung innerhalb des Panels berücksichtigt. 

Wurde eine der Variablen .CURSOR, .CSRPOS, .MSG oder .RESP während Panelverarbeitung auf 
NONBLANK gesetzt, behält sie ihren Wert, bis das aktuelle oder ein anderes Panel erneut aufgeru-
fen wird (ein DISPLAY PANEL(pname) erfolgt), erst dann werden sie wieder zurückgesetzt. 

Die verfügbaren Steuervariablen und ihre Bedeutung folgen im Anschluss. Die syntaktischen Varia-
tionen sind im Syntaxteil nachzulesen. 

Steuervariable Bedeutung 

.ALARM Erzeugt Alarmton 

.ATTR Temporäre Änderung des Feldattributs 

.ATTRCHAR Temporäre Änderung des Attributzeichens 

.AUTOSEL Steuert die Auswahl von Tabellensätzen 

.CSRPOS Cursorpositionierung innerhalb eines Feldes 

.CSRROW Positionierung des CRP auf eine Tabellenzeile 

.CURSOR Cursorpositionierung auf ein Feld 

.HELP Benennung eines Help-Panels 

.MSG Definiert die nächste anzuzeigende Nachricht 

.PFKEY Enthält die gedrückte PF-Taste 

.RESP Ereignistaste (ENTER, END, etc.) 

.TRAIL Rest einer Zeichenfolge nach dem Beschneiden 

.TYPE Nachrichtentyp 

.ZVARS Zuweisung der Platzhalter &Z im Panel 

4.16.4. Systemvariablen 

Der besseren Übersicht wegen sind die Systemvariablen zu logischen Gruppen zusammengefasst. 
Der Aufbau der Tabellen sieht folgendermaßen aus: 

Spalte 1 Name der Systemvariablen 
Spalte 2 Benennt den Pool, in dem sich die Variable befindet. 

F Function Pool 
S Shared Pool 
P Profile Pool 
A in irgendeinem der Pools 

Spalte 3 Variablentyp. Hier sind folgende Abkürzungen möglich: 
E Eingabevariable 
A Ausgabevariable 
EA Ein-/Ausgabevariable 
UV Unveränderbare Ausgabevariable 

Spalte 4 Länge der Variablen in Byte 
Spalte 5 Beschreibung der Variablen 

4.16.4.1. Datum und Zeit 

Name Pool Typ Länge Beschreibung 

ZDATE S UV 8 Tagesdatum, Format sprachabhängig 

ZDATEF S UV 8 Tagesdatum: DD/MM/YY 

ZDATEFD S UV 8 Wie ZDATEF aber sprachabhängig 

ZJDATE S UV 6 Tagesdatum julianisch: YY.DDD 

ZDAY S UV 2 Tagesdatum: DD 

ZMONTH S UV 2 Tagesdatum: MM 

ZTIME S UV 5 Uhrzeit: HH:MM 

ZYEAR S UV 2 Tagesdatum: YY 

4.16.4.2. Generelle Informationen 

Name Pool Typ Länge Beschreibung 

Z S UV 0 Platzhalter 

ZAPLCNT S UV 4 Zahl Aufrufe von APL in einem Screen 

ZAPPLID S UV 8 Applikationskennung 

ZENVIR S UV 32 Umgebungsbeschreibung 
1-8 = Produktname 
9-16 = Betriebssystem 
17-24 = Systemumgebung 
25-32 = Blanks (reserviert) 

ZISPFRC S E 8 Return Code von ISPSTART 

ZLANG P UV 8 Dialogsprache (installationsabhängig) 

ZLOGON S UV 8 Name der Logon-Prozedur 

ZPLACE P EA 7 Lage der Befehlszeile (ASIS/BOTTOM) 

ZPREFIX S UC 8 Prefix aus dem TSO-Benutzerprofil 

ZPROFAPP P E 8 Name des Profile Pools 

ZRXMSG S EA 8 Message-ID aus Panel-REXX 

ZRXRC S EA  Return Code aus Panel-REXX 

ZTEMPF S UV 8 Temporäre Ausgabe  File Tailoring 

ZTEMPN S UV 8 DD-Name von ZTEMPF 

ZUSER S UV 8 TSO-Benutzerkennung 

ZVERB S A 8 SetVerb-Instruktion aus CMD-Table 

ZWINTTL F E  Titelzeile für PopUp-Windows 

4.16.4.3. Bildschirm und PF-Tasten 

Name Pool Typ Länge Beschreibung 

Z S UV 0 Platzhalter 

ZCOLOR S UV 4 Anzahl möglicher Farben (1 oder 7) 

ZDBCS S AV 3 DBCS-Fähigkeit des Terminals (YES/NO) 

ZHILITE S UV 3 Hilite möglich (YES/NO) 

ZKEYS P A 4 Anzahl PF-Tasten 

ZPFCTL P EA 5 PFSHOW-Befehl möglich (YES/NO) 

ZPFFMT P EA 4 Anzahl PF-Tastenanzeige je Zeile 

ZPFSET P EA 4 Angezeigtes PFK-Tasten-Set 

ZPFSHOW P EA 4 Befehlsstatus PFSHOW 

ZPFxx P EA <=255 PF-Tastenbelegung 

ZPFLxx P EA 8 PFK-Label-Inhalt 

ZSCREEN S UV 2 Logischer Bildschirm 

ZSCREENI S UV  Bildschirmpuffer 

ZSCREENC S UV  Cursorposition in ZSCREENI (Achtung, es ist ein Offset, 
für die Byteposition immer +1 

ZSCREEND S UV 4 Bildschirmtiefe (Anzahl Zeilen) 

ZSCREENW S UV 4 Bildschirmbreite (Anzahl Spalten) 

ZSCRMAXD S UV 4 Maximale Bildschirmtiefe im Dialog 

ZSCRMAXW S UV 4 Maximale Bildschirmbreite im Dialog 

ZSPLIT S UV 3 Split Screen aktiv (YES/NO) 

ZTERM P A 8 Terminaltyp wie in 0.1 definiert 

4.16.4.3.1. Scrolling 

Name Pool Typ Länge Beschreibung 

ZSCBR P EA 4 Scrollamount für Browse 

ZSCED P EA 4 Scrollamount für Edit 

ZSCML P EA 4 Scrollamount für Memberliste 

ZSCROLLA S A 4 Wert aus dem Scrollamount-Feld 

ZSCROLLN S A 4 Scrollnumber errechnet aus ZSCROLLA 

ZSCROLLD A E 4 Standardwert für Scroll dynamischer  Bereiche und Tabel-
len 

4.16.4.3.2. Table Display Service 

Name Pool Typ Länge Beschreibung 

ZTDMARK F A 4 Ersatzweise zu "Bottom of Data" 

ZTDMSG A E 8 Benutzerdefinierter TOP/DOWN-Indikator 

ZTDROWS F A 6 Ausgewählte Sätze nach TBDISPL 

ZTDLROWS F A 6 Sätze in der logischen Tabelle 

ZTDSELS F A 4 Ausgewählte Sätze nach TBDISPL 

ZTDTOP F A 6 CRP des obersten sichtbaren Satzes 

ZTDLTOP F A 6 Position der TOP-Zeile der physischen Tabelle in Relation 
zur Position in der logischen Tabelle 

ZTDRET F E 8 Bestimmt, ob Scroll-Return benutzt wird 

ZTDADD F A 3 Mehr Sätze für Scrolling nötig (YES/NO) 

ZTDSCRP F EA 6 CRP der TOP-Zeile nach Scrollvorgang 

ZTDSRID F A 6 RowID des Satzes, auf den ZTDSCRP zeigt 

ZTDAMT F A 4 Anzahl Sätze, die für Scrolling nötig sind 

 

4.17. ISPF Beispiele 

4.17.1. ISPF-DD-Zuweisungen 

Die Zuweisungen für Panel-Bibliotheken und dergleichen zu ermitteln ist nicht so simpel, da die 
Erweiterungen über LIBDEF diese unter dynamischen DD-Namen durchführen. Über den QLIBDEF-
Service ist dies aber möglich, wie das folgende Beispiel zeigt 

/* REXX 

*/ 

parse source . RxTyp RxName . 

 

arg dd 

if dd = "" then , 

   dd="ISPILIB ISPLLIB ISPPLIB ISPMLIB ISPSLIB ISPFILE ISPTLIB ISPTABL" 

Info = "" 

address ispexec 

do § = 1 to words(dd) 

   type = 

   id   = 

   ddn= word(dd,§) 

   "qlibdef" ddn "type(type) id(id)" 

   Info = strip(Info ddn type id"~") 

End 

if RxTyp <> "COMMAND" then return Info 

do while Info <> "" 

   parse var info ddn typ id "~" info 

   say left(ddn,8) typ 

   id = Strip(translate(id,"  ","',")) 

   do §=1 to words(id) 

      say left("",8) ">" word(id,§) 

   end 

end 

exit 

/* Programmbeschreibung ********************************************** 

 

   Ermittelt die Dateinamen von mittels LIBDEF dynamisch zugeweisenen 

   DD-Namen 

 

-- Update History ---------------------------------------------------- 

 

   29.5.2015 Wittemann 

             ° Ausgabe formatiert 

             ° Jetzt auch als Function und Subroutine 

*/ 

4.17.2. Enqueue auf Datei 

/* REXX 

   Beschreibung und History am Ende 

*/ 

parse upper Arg EQDSN 

parse source . rxTYP rxNAME . 

call MSG "OFF" 

if rxTYP <> "COMMAND" & EQDSN = "" then return "Keine Datei angegeben!" 

 

call check_enqueues 

 

if rxTYP <> "COMMAND" then return result 

 

Queue 

call bpxwdyn "alloc lrecl("max") space(1 75) tracks rtddn(tempdd)" 

address tso "EXECIO * DISKW" tempdd "(FINIS" 

ADDRESS ISPEXEC "VIEW DATASET('&tempdsn')" 

Exit 

 

Check_enqueues: 

   address ispexec 

   "CONTROL ERRORS RETURN"       /* Ensure that errors not cancel REXX */ 

   "TBEND ENQ"                   /* Destroy table from QUERYENQ        */ 

   EQDSN = Strip(EQDSN,,"'")     /* Strip off quotes if DSN quoted     */ 

   MAJOR = '*'                   /* Set Major to any Name              */ 

   MINOR = EQDSN'*'              /* Set Minor to Arg DSN w/asterisk    */ 

   "QUERYENQ TABLE(ENQ) QNAME(MAJOR) RNAME(MINOR) REQ(*) XSYS LIMIT(0)" 

   QRC = RC                      /* Save RC from QUERYENQ              */ 

   If QRC < 5 Then Do            /* If 0-2-4, an ENQ is being held     */ 

   "TBSTATS DSNENQ ROWCURR(nrows)" 

   ret_erg = 

   if rxTYP = "COMMAND" then do 

      queue   "Holder---How---Type---Scope---System---Qname----" !! , 

            "Rname---------------------------------------Member--" 

   End 

   "TBTOP DSNENQ" 

   DO nrows 

      "TBSKIP ENQ" 

      if rxTYP = "COMMAND" then , 

         queue   LEFT(zenjob,9)    !! , 

                 LEFT(zendisp,6)   !! , 

                 LEFT(zenhold,7)   !! , 

                 LEFT(zenscope,8)  !! , 

                 LEFT(zensyst,9)   !! , 

                 LEFT(zenqname,9)  !! , 

                 LEFT(zenrname,60) !! , 

                 "" 

      else , /* Subroutine oder Function */ 

         ret_erg = strip(ret_erg  , 

               strip(zenjob   ) , 

               strip(zendisp  ) , 

               strip(zenhold  ) , 

               strip(zenscope ) , 

               strip(zensyst  ) , 

               strip(zenqname ) , 

               strip(zenrname )"°") 

   END 

   "TBEND DSNENQ"         /* Destroy table from QUERYENQ      */ 

   if rxTYP = "COMMAND" & queued() < 2 then qrc = 8 

End 

If QRC = 8 Then Do                   /* If 8, No ENQ held       */ 

   if rxTYP = "COMMAND" then do 

      queue   "Keine ENQUEUES auf Datei" EQDSN 

      queue   " " 

   end 

End 

return ret_erg 

 

/* Programmbeschreibung ********************************************** 

 

   Ermittlung der Enqueues auf Datei(en) 

   By Jim Moore - Concentrated Logic Inc - 1/15/2005 

 

-- Update History ---------------------------------------------------- 

   20.12.2009 Wittemann 

              Keine Tabellenanzeige mit Panel 

              Information wird in Datei geschrieben und mit VIEW gezeigt 

              Wahlweise als COMMAND, SUBROUTINE oder FUNCTION 

*/ 

4.17.3. Enqueue auf Member 

/* REXX * Funktion zur Ermittlung von Enqueues auf PO-Member 

*/ 

parse source . rxTYP rxNAME . 

address isredit "Macro (parm)" 

if RC = 0 then do 

   if parm <> "" then parse var parm eqdsn "(" member ")" 

   else do 

      address isredit "(EQDSN) = dataset" 

      address isredit "(MEMBER) = member" 

   end 

End 

else parse upper value strip(arg(1),"B","'") with eqdsn "(" member ")" 

if member = "" then call abend "Kein Member zur ENQ-Prüfung vorhanden." 

erg = 

address ispexec 

"CONTROL ERRORS RETURN" 

"TBEND DSNENQ"                   /* Destroy table from QUERYENQ      */ 

MAJOR = '*'                      /* Set Major to any Name            */ 

MINOR = EQDSN'*'                 /* Set Minor to Arg DSN w/asterisk  */ 

 

"QUERYENQ TABLE(DSNENQ) QNAME(MAJOR) RNAME(MINOR) REQ(*) XSYS LIMIT(0)" 

If RC < 5 Then Do               /* If 0-2-4, an ENQ is being held   */ 

   "TBSTATS DSNENQ ROWCURR(nrows)" 

   "TBQUERY DSNENQ KEYS(keys) NAMES(names)" /* Felder ermitteln */ 

   parse var keys  . "(" keys ")" . 

   parse var names . "(" names ")" . 

   fldNAMES = strip(keys names) 

   "TBTOP DSNENQ" 

   "TBSKIP DSNENQ" 

   Do while RC = 0 

      tmp = 

      do i=1 to words(fldNAMES) 

         feld=word(fldNAMES,i) 

         inhalt = value(feld) 

         tmp = strip(tmp !! feld"='"inhalt"';") 

      end 

      if pos(member,tmp) > 0 & pos(eqdsn,tmp) > 0 then do 

         erg = zenjob zensyst 

         leave 

      end 

      "TBSKIP DSNENQ" 

   End 

   "TBEND DSNENQ" 

End 

 

if rxTYP <> "COMMAND" then return erg 

if erg = "" then erg = "Kein ENQUEUE auf" eqdsn"("member")." 

say erg 

exit 

 

abend: zedlmsg = arg(1);address ispexec "setmsg msg(isrz000)";exit 

 

/* History ------------------------------------------------------ 

   By Jim Moore - Concentrated Logic Inc --- 01/15/2005 

   By Mike Tomkins - DaimlerChrysler Corp. - 01/20/2006 

   -------------------------------------------------------------- 

   Updates 

   17.09.2009 Wittemann 

              ° Alle Felder der ENQ-Table ermitteln und 

                Blockade auf Member-Ebene feststellen 

              ° Läuft auch als Edit Makro 

              ° dsn(mbr) kann auch bei EditMakro übergeben werden 

*/ 

4.17.4. Anzeige von Member-Generationen in Browse, Edit oder View 

Leider liefert die IBM im Edit oder View keinerlei vernünftige Unterstützung für die Anzeige von Mem-
ber Generations. Man bekommt es irgendwie hin, aber der Weg ist sehr steinig. Vielleicht liegt es 
daran, dass man gerne das Produkt "Dataset Manager" für Geld an den Mann bringen möchte. Ein 
Schelm, wer Böses dabei denkt. 

Das Beispiel zeigt eine Lösung, welche im Edit oder View als Macro gestartet werden kann, aber 
auch im Browse als Command aufrufbar ist. Die Anzeige erfolgt, um konkurrierende Edit-Zugriffe zu 
vermeiden, im View. Dies ist aber vollkommend ausreichend um beispielsweise der Vorgeneration 
auf die aktuelle durchzuführen, oder entsprechend unter anderem Namen abzuspeichern oder… 

/* REXX 

   Programmbeschreibung und Update History am Ende 

*/ 

call msg "OFF" 

parse source . . RxName . 

address ispexec 

"control errors return" 

address isredit "macro (GEN)" 

macro = RC = 0 

if ^Macro then do 

   GEN = arg(1) 

   "vget zscreeni" 

   zscreeni = space(zscreeni,1) 

   BrowseFound = pos(" BROWSE ",left(zscreeni,240)) 

   if BrowseFound = 0 then call abend , 

      "Als Command nur in einem Member im Browse möglich" 

   parse var zscreeni . =(BrowseFound) . DS "(" MEM ")" . 

end 

else do 

   address isredit "(DS)  = dataset" 

   address isredit "(MEM) = member" 

end 

ds = strip(ds) 

if MEM = "" then call abend "läuft nur in einem Member" 

"dsinfo dataset('&ds')" 

if zdsngen < 1 then call abend "Nur in PO/E-2 mit Generations möglich" 

if Gen = "" then Gen = -1 

if ^datatype(gen,"W") ! gen >= 0 then , 

   call abend "Keine gültige Vorversion angegeben (-1, -2...)" 

"view dataset('&ds(&mem)') gen(&gen)" 

Select 

   when RC = 0  then nop 

   when RC = 10 then call abend "Generation" gen "nicht vorhanden." 

   when RC = 14 then call abend MEM "in use by another user" 

   otherwise         call abend "Unerwarteter RC:" rc 

end 

exit 

 

abend: 

   zedlmsg=RxName"-Error:" arg(1) 

   address ispexec "setmsg msg(isrz000)" 

exit 

 

/* Programmbeschreibung *********************************************** 

 

   Anzeige einer Vorgeneration eines Members. 

 

   Innerhalb einer Memberanzeige im EDIT oder VIEW läuft das Programm 

   als Edit Macro. 

 

   In einer Memberanzeige im Browse, kann es als Command aufgerufen 

   werden. In diesem Fall wird Datei- und Membername aus ZSCREENI 

   ermittelt. 

 

   Erfolgt der Aufruf nicht aus einem Member einer PO/E-Type2 

   Bibliothek oder wurden keine Genereations angelegt, oder die 

   angegebene Generation ist nicht vorhanden, wird ein entwprechender 

   Hinweis ausgegeben. 

 

   Parameter: Relative Generationsnummer (-1, -2, etc.) 

              Wird keine Generation übergeben, wird -1 angenommen 

 

-- Update History ----------------------------------------------------  

 

   Datum_____ UserID  Name____________________________________________  

 

   03.05.2021 SP1     Wittemann, Karlheinz 

*/ 

4.17.5. Variablen aus Datei einlesen 

Um den Pflegeaufwand einer Anwendung zu minimieren, werden sehr oft Inhalte für Variablen in 
einer Parameterdatei abgelegt, die einfach zu editieren sind.  

Das folgende Beispiel zeigt ein Programm, das daten aus einer Datei liest und interpretativ zu Vari-
ablen mit Inhalten umsetzt. Dabei werden alle Variablen, welche die Namensregeln für ISPF erfüllen, 
in den SHARED oder PROFILE Pool geschrieben. Die Namen aller ISPF-Variablen werden in einer 
Variablen abgelegt, deren Name als Parameter mitgegeben werden kann. Welches Zeichen als Se-
parator oder Kommentar bei der Umsetzung berücksichtigt werden soll, ist ebenfalls steuerbar. 

Alle anderen werden zu einem interpretierbaren String zusammengesetzt, den der Aufrufer über den 
INTERPRET-Befehl umsetzen kann. Ist ISPF nicht aktiv, werden alle Variablen so behandelt. 

In der Variablendefinition der eingelesenen Sätze können auch Ausdrücke oder Funktionsaufrufe 
vorkommen, welche ebenfalls berücksichtigt werden. 

/* REXX * getEvars 

   Programmbeschreibung und Update-History am Programmende 

*/ 

parse source . rxTYPE rxNAME . 

if rxTYPE = "FUNCTION" then arg varDSN,POOL,SEP,TOTALVAR,DUMMY 

else                        arg varDSN POOL SEP TOTALVAR . 

 

call Check_VarDsn 

 

if SEP      = "" then SEP      = "*" 

if TOTALVAR = "" then TOTALVAR = "ALLVARS" 

 

call msg "OFF" 

"alloc dd(extvars) da('"vardsn"') shr reuse" 

if rc <> 0 then return , 

   "errCODE="RC";errTEXT='Fehler bei ALLOC" varDSN "in" rxNAME"';" 

"execio * diskr extvars (stem evars. finis" 

if rc <> 0 then return , 

   "errCODE="rc";errTEXT='Fehler bei EXECIO" varDSN "in" rxNAME"';" 

if evars.0 < 1 then return , 

   "errCODE=99;errTEXT='Keine Sätze in" varDSN "vorhanden ("rxNAME")';" 

"free dd(extvars)" 

 

ispPOOL  = (wordpos(POOL,"SHARED PROFILE") > 0) 

"ISPQRY" 

ispACTIV = (RC = 0) 

ispVARS  = 

prgVARS  = 

prgERGS  =  

if ispACTIV then "ISPEXEC VGET (ZSCREEN,ZAPPLID) SHARED" 

 

do §=1 to evars.0 

   if word(evars.§,2) = "=" then evars.§ = delword(evars.§,2,1) 

   evars.§ = space(strip(evars.§)) 

   if evars.§ = "" then iterate 

   if left(evars.§,length(SEP)) = SEP then iterate 

   parse var evars.§ varNAME varVALUE (SEP) 

   do while pos("[",varVALUE) > 0 

      parse var varVALUE _1 "[" varsub "]" _2 

      interpret "value = " varsub 

      varVALUE = strip(_1) !! value !! strip(_2) 

   end 

   isISPvar = ispVARnameOK(varNAME) 

   if wordpos(varNAME,ispVARS) = 0 & isISPvar then do 

      ispVARS = strip(ispVARS varNAME) 

   end 

   if wordpos(varNAME,prgVARS) = 0 & ^isISPvar then do 

      prgVARS = strip(prgVARS varNAME) 

   end 

   interpret varNAME "=" varVALUE 

end 

do §=1 to words(prgVARS) 

   varname = word(prgVARS,§) 

   prgERGS = strip(prgERGS varname"='"value(varname)"';") 

end 

address ispexec 

if words(ispVARS) > 0 then do 

   do §=1 to words(ispVARS) 

      "vput ("word(ispVARS,§)") &pool" 

   End 

   interpret TOTALVAR "= ispVARS" 

   "vput (&totalvar) &pool" 

End 

if rxTYPE <> "COMMAND" then return prgERGS 

do while prgERGS <> "" 

   parse var prgERGS varNAME "=" varVALUE ";" prgERGS 

   say left(strip(varNAME)" ",15,".") strip(strip(varVALUE,,"'")) 

end 

exit 

 

Check_VarDsn: procedure expose VarDsn 

   VarDsn = strip(VarDsn,,"'") 

   if sysdsn("'"VarDsn"'") = "OK" then return 

   parse var VarDsn VarDsn "(" member ")" 

   if length(VarDsn) < 9 then do 

      call listdsi VarDsn "FILE" 

      VarDsn = SysDsName 

      if Member <> "" then VarDsn = VarDsn"("Member")" 

   end 

return 

 

ispVARnameOK: procedure expose ispPOOL ispACTIV 

   /****************************************************************** 

   * Entscheidet, ob die Variable im ISPF-Pool eingestellt wird      * 

   ******************************************************************/ 

   arg varname 

return "" !! , 

   (verify(varname,"ABCDEFGHIJKLMNOPQRSTUVWXYZ0123456789#§$") = 0) & , 

   (pos(left(varname,1),"ABCDEFGHIJKLMNOPQRSTUVWXYZ#§$") > 0     ) & , 

   (length(varname) <= 8) & ISPpool & ispACTIV 

 

/* Programmbeschreibung ********************************************** 

 

   Liest Datensätze und führt Variablenzuweisungen durch. 

   Word-1    ist der Variablenname 

   Subword-2 ist der Wert. Dieser kann Funktionen zwischen [] enthalten 

 

   Die Variablennamen und Werte werden als Stringkette mit ; getrennt 

   aufgebaut und können im rufenden Programm mit INTERPRET umgesetzt 

   werden. 

 

   Per Default werden die Variablen, welche die ISPF-Konventionen 

   erfüllen im Pool abgestellt. Wir kein Pool übergeben, oder ist 

   ISPF nicht aktiv, werden alle Variablen als interpretierbares RESULT 

   zurückgegeben. 

 

   Parm-1  Name der Datei, in der die Externen Variablen liegen 

   Parm-2  Wenn SHARED oder PROFILE, werden alle Variablen, deren Name 

           <= 8 Byte ist, in den entsprechenden ISPF-POOL geschrieben 

 

   (C) sp Service Professional GmbH 2021 

-- Update History ----------------------------------------------------  

 

   Datum_____ UserID  Name____________________________________________  

 

   03.05.2012 SP1     Wittemann, Karlheinz Erstellung 

   24.08.2012 SP1     Wittemann, Karlheinz 

              ° Es kann ein Separator-Zeichen mitgeliefert werden, 

                anhand dessen Kommentare erkannt werden, da der "*" 

                manchmal in den Variablen vorkommt (z.B. bei JCL) 

   24.09.2018 SP1     Wittemann, Karlheinz 

              ° Statt Dateiname kann auch ein DD-Name übergeben werden. 

*/ 

4.17.6. Pflege einer Command Table 

Für das nachfolgende Beispiel gehen wir davon aus, dass in der ISPF-Configurationtabelle drei 
mögliche User-Commandtables vorgesehen und unter USR1|2|3CMDS definiert sind, sowie drei 
mögliche Site Tables (unternehmensspezifisch), welche unter den Namen SYS1|2|3CMDS erwartet 
werden. 

/* REXX 

   Programmbeschreibung und Update History am Programmende 

*/ 

parse source . . RxName . 

 

parse upper arg parms 

address ispexec 

"control errors return" 

Main_Start: 

   call set_vars 

   call search_4_isptlib 

   call check_cmdtab 

   call Show_Table 

Main_End: 

Exit 

 

search_4_isptlib: 

   /****************************************************************** 

   * Alle unter ISPTLIB definierten Dateien werden ermittelt         * 

   * Gesucht wird das Dataset, in dem die Commandtable ggfls bereits * 

   * existiert.                                                      * 

   ******************************************************************/ 

   "qbaselib isptlib id(TLib)" 

   TLib = translate(TLib," ","',") 

   if TLib = "" then call abend "ISPTLIB-Definition fehlt" 

   CTabDsn      =  

   CmdTab_Found = 0 

   do §=1 to words(TLib) 

      _DSN = word(TLib,§) 

      mbrs = lst#mbr(_DSN) /* User Defined Function */ 

      if wordpos(cmdtab,mbrs) > 0 then do 

         CmdTab_Found = 1 

         CTabDsn      = _DSN 

         Leave 

      End 

   End 

   /*************************************************************** 

   * Wurde keine Commandtable gefunden, wird nach der ersten      * 

   * Datei unter ISPTLIB gesucht, welche als HLQ die UserID des   * 

   * Benutzers enthält.                                           * 

   ***************************************************************/ 

   if ^CmdTab_found then do 

      do §=1 to words(TLib) 

         parse value word(TLib,§) with hlq "." 

         if hlq <> userid() then iterate 

         CTabDsn = word(TLib,§) 

         Leave 

      End 

   End 

   /*************************************************************** 

   * Wird auch keine ISPTLIB des Users gefunden, wird die         * 

   * Verarbeitung mit Ausgabe einer Fehlernachricht beendet.      * 

   ***************************************************************/ 

   if ^CmdTab_Found & CTabDsn = "" then , 

      call abend "Command Table nicht gefunden und keine" , 

                 "USER-ISPTLIB-Datei definiert." 

Return 

 

check_cmdtab: 

   /*************************************************************** 

   * Es wird geprüft, ob die Command Table bereits geöffnet ist   * 

   ***************************************************************/ 

   "qtabopen list(otabs.)" 

   TabOpen = 0 

   do §=1 to otabs.0 

      if otabs.§ = cmdtab then do 

         TabOpen = 1 

         Leave 

      End 

   End 

   /*************************************************************** 

   * Wird sie im Speicher nicht gefunden, wird sie angelegt.      * 

   ***************************************************************/ 

   if ^TabOpen then do 

      "tbcreate &cmdtab names(&ctnames) write" 

      "tbsort &cmdtab fields(zctverb,c,a)" 

   End 

Return 

 

DatenOK: 

   /*************************************************************** 

   * Prüfen und Auswerten von ZCMD                                * 

   ***************************************************************/ 

   Delete  = abbrev("DELETE",PCmd,1) 

   Insert  = abbrev("INSERT",PCmd,1) 

   BadCmd  = ^Delete & ^Insert 

   LenOK   = length(parm) > 1 & length(parm) < 9 

   BadParm = datatype(left(Parm,1),"N") 

   Daten_OK = 0 

   Select 

      when PCmd <> "" & BadCmd    then MSG="'"PCmd"' unbekannt!" 

      when PCmd <> "" & Parm = "" then MSG="Kein Befehl übergeben" 

      when PCmd <> "" ^LenOK      then MSG="Befehlslänge 2 bis 8 Byte" 

      when BadParm                then MSG="Stelle 1 nicht CHAR" 

      otherwise Daten_OK = 1 

   end 

   if ^Daten_OK then call Set_Msg MSG 

return Daten_OK 

 

Show_Table: 

   /****************************************************************** 

   * Anzeige und Verarbeitung der User oder Side Command Table       * 

   ******************************************************************/ 

   UpdFlag = 0 

   do forever 

      "TbTop &CmdTab" 

      "TbSkip  &CmdTab number(&CrpTop)" 

      "TbDispl &CmdTab panel(CmdTab)" 

      if abbrev("CANCEL",translate(ZCMD),1) then return 

      if RC > 4 then do 

         if ZTdSels > 0 then call update_Table 

         leave 

      end 

      CrpTop = ZTdTop 

      parse upper var ZCMD PCmd Parm . 

      if ^DatenOK() then iterate 

      /*************************************************************** 

      * Prüfen und Auswerten von ZCMD                                * 

      ***************************************************************/ 

      Select 

         when Delete then do 

            /********************************************************* 

            * Löscht einen Eintrag und positioniert den CRP.         * 

            *********************************************************/ 

            UpdFlag = 1 

            ZCTVERB = Parm 

            "TbTop &CmdTab" 

            "TbScan &CmdTab arglist(ZCTVERB) condlist(EQ)" 

            if RC = 0 then do 

               "TbDelete &CmdTab" 

               if CrpTop > 1 then CrpTop = CrpTop – 1 

            end 

            if rc = 8 then call Set_Msg ZCTVERB "nicht vorhanden" 

         end 

         when Insert then do 

            /********************************************************* 

            * Neuer Eintrag. Tabellenfelder werden vorbelegt. Vorab  * 

            * wird geprüft, ob es einen gleichnamigen Eintrag gibt.  * 

            *********************************************************/ 

            UpdFlag = 1 

            "TbTop &CmdTab" 

            ZCTVERB = Parm 

            ZCTTRUNC = "0" 

            ZCTACT   = "NOP" 

            ZCTDESC  = "Kommentar einfügen!" 

            "TbScan &CmdTab arglist(ZCTVERB) condlist(EQ)" 

            if RC > 0 then do 

               "TbAdd &CmdTab order" 

               "TbGet &CmdTab position(CrpTop)" 

               Iterate 

            End 

            call Set_Msg ZCTVERB "existiert bereits!" 

         end 

         otherwise call Update_Table 

      end 

   end 

   call save_table 

return 

 

Update_Table: 

   /***************************************************************** 

   * Update der Tabellensätze.                                      * 

   *****************************************************************/ 

   UpdFlag = 1 

   do while ZTdSels > 0 

      "TbPut &CmdTab order" 

      if ZTdSels > 1 then "TbDispl &CmdTab" 

      else ZTdSels = 0 

   end 

return 

 

Save_Table: 

   /****************************************************************** 

   * Bei Programmende werden die Änderungen der Command Table auf    * 

   * Platte zurückgeschrieben, sofern es mindestens eine Änderung    * 

   * gab und die Tabellenansicht nicht mit CANCEL beendet wurde.     * 

   * Für das Zurückschreiben wird eine Temp-Tabelle benutzt, da das  * 

   * Original nicht direck verändert werden kann (die Command Table  * 

   * wird bei Start von ISPF mit NOWRITE geöffnet). Der Name der     * 

   * Temp-Tabelle wird dynamisch erzeugt und ist UNIQUE.             * 

   * Die Command Table wird auf die Temp Table kopiert und vorüber-  * 

   * gehend in die Datei der Commandtable geschrieben. Danach wird   * 

   * das Member auf den originären Namen der Commandtable umbenannt. * 

   ******************************************************************/ 

   if ^UpdFlag then return 

   do forever 

      TempTab = "T"right(time("L"),6) 

      if sysdsn("'"CTabDsn"("TempTab")'")="MEMBER NOT FOUND" then leave 

   end 

   "libdef isptabl dataset id('&CTabDsn') stkadd" 

   "tbcreate &temptab names(&CTnames) write" 

   "tbsort   &temptab fields(zctverb,c,a)" 

   "tbtop &CmdTab" 

   "tbskip &CmdTab" 

   do while rc = 0 

      "tbadd &temptab order" 

      "tbskip &CmdTab" 

   End 

   "TbClose &TempTab" 

   "libdef isptabl" 

   "lminit dataid(did) dataset('&CTabDsn') enq(SHRW)" 

   "lmopen dataid(&did) option(output)" 

   "lmmren dataid(&did) member(&TempTab) newname(&CmdTab)" 

   "lmclose dataid(&did)" 

   "lmfree dataid(&did)" 

Return 

 

set_vars: 

   /****************************************************************** 

   * Interpretation der Übergabe-Parameter                           * 

   ******************************************************************/ 

   Type = "USER" 

   do §=1 to words(parms) 

      if datatype(word(parms,§),"W") then CmdTbNr = word(parms,§) 

      if abbrev("SITE",word(parms,§),1) then Type = "SITE" 

   end 

   /****************************************************************** 

   * Allgemeine Variablen                                            * 

   ******************************************************************/ 

   if wordpos(CmdTbNr,"1 2 3") = 0 then CmdTbNr = "1" 

   Title.USER = "Pflege der User Command Table" 

   Title.SITE = "Pflege der Site Command Table" 

   Table.USER = "USR"CmdTbNr"CMDS" 

   Table.SITE = "SYS"CmdTbNr"CMDS" 

 

   CmdTab    = Table.Type 

 

   Titel     = "^"center("<°"Title.Type" ("CmdTab")^>",79,"-") 

   CrpTop    = 1 

 

   CTNames   = "ZCTVERB ZCTTRUNC ZCTACT ZCTDESC" 

   ZCTTRUNC  = "0" 

   ZCTACT    = "NOP" 

   ZCTDESC   = "Kommentar einfügen!" 

Return 

 

AbEnd:   parse arg zedlmsg;"setmsg msg(isrz000)";exit 

Set_Msg: parse arg zedlmsg;"setmsg msg(isrz000)";return 

 

/* Programmbeschreibung ************************************************ 

 

   Anlegen/Pflegen einer Command Table. 

 

   Parameter: 

      SITE   Pflegt die SITE-Table >> sonst die USER-Table 

      1|2|3  Benennt die jeweilige User Table (1=Default) 

             oder die entsprechende Site-Table, deren Namensprefix 

             in der ISPF-Config-Table festgelegt wird. Wir unterstellen 

             hier dass in der ISPF Config Table je 3 Commandtables 

             möglich sind und für die User Table als Präfix USR, für 

             die Site Table SYS definiert ist. Daraus ergeben sich 

             Membernamen unter ISPTLIB wie USR1CMDS oder SYS2CMDS. 

 

   Commands: 

      <I>nsert Verbname erstellt ein Eintrag 

      <D>elete Verbname löscht ein Eintrag 

      <C>ancel Programmende ohne Datensicherung 

 

   Sofern die Tabelle vorhanden ist, werden Änderungen im gleichen 

   Dataset durchgeführt. Generell könnte dies zur Folge haben, dass 

   der Anwender dort kein Schreibrecht hat. Dies wird billigend in 

   Kauf genommen. Vorteil: Eine User-Table kann auch mehreren Personen 

   oder Abteilungen zur Verfügung gestellt werden. Updates könnten 

   dann gegebenenfalls nur einige Benutzer durchführen. 

 

-- Update History ------------------------------------------------------ 

 

   02.02.2016 Zitzmann, Erstellt 

 

   03.05.2021 Wittemann, Karlheinz 

              ° Tabellenhandling überarbeitet, Änderungen ziehen sofort 

*/ 

Und hier das dazugehörende Tabellenpanel: 

)attr default(%+_) 

    ] type(output)  color(red) 

    [ type(input)   color(green) caps('ON') 

    ² type(input)   color(turq) 

    ^ type(dataout) color(blue) 

    ° type(dataout) color(white) 

    µ type(Text)    color(blue)  hilite(uscore) 

    ³ area(dynamic) 

    { type(char)    color(white) hilite(uscore) 

    } type(char)    color(blue) 

    ~ type(char)    color(turq) 

)body expand(\\) 

³Titel                                                                  ³ 

%COMMAND ==>_ZCMD\ \%SCROLL ==>_SAMT 

% 

+PrimCMDs: ³FuncVar,ShadVar                                             ³ 

% 

+Befehl    T   Aktion 

µ        µ    µBeschreibung 

+ 

)model 

]z        [z% [z 

%             ²z 

)init 

  .zvars = '(zctverb zcttrunc zctact zctdesc)' 

  &FuncVar = 'Insert befehl  Delete befehl  Cancel' 

  &ShadVar = '{}}}}} ~~~~~~  {}}}}} ~~~~~~  {}}}}}' 

)proc 

   ver(&zcttrunc,NB,RANGE,0,8) 

   ver(&zctact,NB) 

   ver(&zctdesc,NB,LEN,'>',10) 

)Field 

   Field(zctact)   len('100') 

)end 

/* 

/* Update History 

/* 

/* 08.02.2016 Zitzmann 

/*       ° Primärbefehl CANCEL ergänzt 

/* 

/* 06.07.2016 Wittemann 

/*       ° Dynamische Titelzeile, weil auch Site-Table möglich ist 

/* 

/* 24.03.2017 Zitzmann 

/*       ° ZCTACT ist nun scrollable. 

4.17.7. Eine kleine ISPF-Anwendung 

Die nachfolgende Anwendung ist eine Aufgabe, welche in diversen ISPF-Seminaren zu lösen war. 
Darin kommen die meisten Alltagsprobleme in unterschiedlichen Variationen vor. Bei der Aufgaben-
stellung sollten folgende Punkte erfüllt werden: 

▪ Arbeiten mit mehreren Tabellen zu einer Zeit 
▪ Nutzung unterschiedlicher Paneltypen (Daten-, Auswahl- und Tabellenpanels) 
▪ Panels als Windows verarbeiten 
▪ Dynamic Areas 
▪ Dynamic Modelsets 
▪ Tabellenverarbeitung über Skeletons und REXX-Logik 

Weil jedes Kind einen Namen braucht nennen wir diese Anwendung AKUVIGG (Auftrags Und Kun-
denVerwaltung Inter Galaktisch Gut) 

Die Beschreibung erfolgt entsprechend der Hierarchie der Reihe nach. Zum jeweiligen Abschnitt 
werden immer erst die Programme, danach die Panels, danach die Skeletons angeführt. 

4.17.8. Die Programme 

4.17.8.1. AKUVIGG – Das Startprogramm 

Das hierarchisch oberste Programm sorgt zunächst für den Aufbau der richtigen Umgebung. Es 
gewährleistet, dass die Anwendung unter der gewünschten Applikationskennung laufen wird. 

Für die Zuweisung der benötigten Bibliotheken für REXXe, Panels und Skeletons wird der Name der 
Bibliothek ermittelt, aus der das AKUVIGG per TSO-EXEC-Befehl gestartet wurde. Von diesem Na-
men werden die Namen der Panel-, Skeleton- und Tabellenbibliotheken abgeleitet. 

Alle Globalen Variablen werden im Startprogramm definiert und in den ISPF-Pools abgelegt. Auf 
diese Weise können alle Unterprogramme (auch die noch nicht geschriebenen) sehr einfach auf 
deren Werte zugreifen. 

/* REXX 

   Programmbeschreibung am Ende 

*/ 

 

parse source . 5 RxTyp 6 . RxName . RxLib . 

if RxTyp <> "C" then call abend RxName "muss als Command laufen" 

address ispexec 

"vget (zapplid zacctnum) shared" 

 

if zapplid <> "AKUV" then do 

   call Check_Command_Table 

   "select cmd(EX '"RxLib"(AKUVIGG)' EXEC)" , 

      "newappl(AKUV) scrname(AKUVIGG) passlib" 

   exit 

end 

 

Main_Start: 

 

   call Set_Vars 

   call Set_Environment 

   call open_tables 

   "select panel(AKUVIGG)" 

   if rc = 4 & Admin then call end_tables 

   else call close_tables 

   call Reset_Environment 

 

Main_Ende: 

exit 

 

Set_Vars: 

   /****************************************************************** 

   * Definition der Bibliotheks-Umgebung                             * 

   ******************************************************************/ 

   ApplHlq  = left(RxLib,lastpos(".",RxLib)) 

   RexxLib  = "'"RxLib"'" 

   PanLib   = "'"ApplHlq"PANELS'" 

   SklLib   = "'"ApplHlq"SKELS'" 

   MsgLib   = "'"ApplHlq"MSGS'" 

   TblLib   = "'"ApplHlq"TABLES'" 

   /****************************************************************** 

   * Felder Kundentabelle                                            * 

   ******************************************************************/ 

   KdKeys   = strip(space(, /* Schlüsselfelder       */ 

              "#kdnr    " , /* Kundennummer          */ 

              ,1)) 

   KdNames  = strip(space(, /* Namensfelder          */ 

              "#kdgr    " , /* Kundengruppe          */ 

              "#nname   " , /* Nachname              */ 

              "#vname   " , /* Vorname               */ 

              "#plz     " , /* Postleitzahl          */ 

              "#ort     " , /* Ort                   */ 

              "#str     " , /* Straße                */ 

              "#telfn   " , /* Telefon Festnetz      */ 

              "#telmob  " , /* Telefon mobil         */ 

              "#fax     " , /* Faxnummer             */ 

              "#gebdat  " , /* Geburtsdatum          */ 

              "#nnameu  " , /* Name Uppercase        */ 

              "#vnameu  " , /* Vorname Uppercase     */ 

              "#ortu    " , /* Ort Uppercase         */ 

              "#stru    " , /* Straße Uppercase      */ 

              "#gebdats " , /* Geburtsdatum jjjjmmtt */ 

              ,1)) 

   /****************************************************************** 

   * Felder der Auftragstabelle                                      * 

   ******************************************************************/ 

   AufNames = strip(space(, /* Namensfelder          */ 

              "$AufKDNR " , /* Auftrags-Kundennummer */ 

              "$ArtANZ  " , /* ArtikelAnzahl         */ 

              "$ArtBEZ  " , /* Artikelbezeichnung    */ 

              "$ArtNUM  " , /* Artikelnummer         */ 

              "$ArtEPR  " , /* Artikel Einzelpreis   */ 

              "$AufVKZ  " , /* Verarbeitungs-KZ      */ 

              "$AufDAT  " , /* Auftragsdatum         */ 

              "$AufDATS " , /* Auftragsdatum jjjjmmtt*/ 

              ,1)) 

   /****************************************************************** 

   * Allgemeine Variablen                                            * 

   ******************************************************************/ 

   Admins   = strip(space(,    /* User mit speziellen Privilegen  */ 

              "XYZ0815"   ,    /* Aloisius Bröselmeier            */ 

              "SP1"       ,    /* Der liebe Heinzi                */ 

              ,1)) 

   Admin    = (wordpos(userid(),Admins) > 0) 

   Heute    = translate("Tt.Mm.CcJj",date("S"),"CcJjMmTt") 

   Absender = "Betrugs-GmbH & Co KG, Lügengasse 12, 00815 Münchhausen" 

   BkHeader = "Institut------------- IBAN-----------------------" 

   BkVerb1  = "Parkbank Zoo          DE42 1234 5678 0000 7738 00" 

   BkVerb2  = "Samenbank Kalsruh     DE73 9876 5432 0143 5548 07" 

   MWSt     = 19 

   Ziel     = 3 

   ActNum   = word(strip(translate(zacctnum," ","(),")),1) 

   KdTab    = "K"reverse(userid()) 

   AufTab   = "A"reverse(userid()) 

   if Admin then pendtext = "Ende ohne Tabellen-Save" 

   else          pendtext = "H a b e   f e r t i g" 

   ShrVars  = Strip(Space(,     /* Variablen für den Shared-Pool   */ 

              "Heute   " ,     /* Tagesdatum im Format tt.mm.jjjj */ 

              "Admin   " ,     /* 0|1 ob Administrator            */ 

              "ActNum  " ,     /* Accountnummer aus LOGON         */ 

              "KdTab   " ,     /* Name der Kundentabelle          */ 

              "AufTab  " ,     /* Name der Auftragstabelle        */ 

              "PEndText" ,     /* Dyn. Text für Admin = 0|1       */ 

              "Absender" ,     /* Für den Rechnungskopf           */ 

              "BkHeader" ,     /* Überschrift Bankverbindungen    */ 

              "BkVerb1 " ,     /* Bankverbindung-1 für Rechnung   */ 

              "BkVerb2 " ,     /* Bankverbindung-2 für Rechnung   */ 

              "MWSt    " ,     /* Mehrwertsteuer                  */ 

              "Ziel    " ,     /* Zahlungsziel in Tagen           */ 

              ,1)) 

   "vput (ShrVars) shared"     /* Namen der Shr-Pool-Variablen    */ 

   "vput (&ShrVars) shared"    /* Inhalte der Shr-Pool-Variablen  */ 

 

   ProfVars = strip(space(,    /* Variablen für den Profile-Pool  */ 

              "ProfVars" ,     /* Namen der Prof-Pool-Variablen   */ 

              "LstRechN" ,     /* Letzte Rechnungsnummer          */ 

              "LstKdNr " ,     /* Letzte Kundennummer             */ 

              "SortArgK" ,     /* Sortierkriterium Kunde          */ 

              "SortArgA" ,     /* Sortierkriterium Auftrag        */ 

              "ASicht  " ,     /* Hinweis Ansicht Auftragstabelle */ 

              ,1)) 

   "vput ProfVars profile" 

   if LstRechN = "" then do 

      LstRechN = 0 

      "vput LstRechN profile" 

   end 

   if LstKdNr  = "" then do 

      LstKdNr  = 0 

      "vput LstKdNr profile" 

   end 

   if SortArgK = "" then do 

      SortArgK = "N" 

      "vput SortArgK profile" 

   end 

   if SortArgA = "" then do 

      SortArgA = "KDN" 

      "vput SortArgA profile" 

   end 

   if ASicht   = "" then do 

      ASicht   = 1 

      "vput ASicht profile" 

   end 

   /****************************************************************** 

   * Headerlines für das 1. Selection-Panel                          * 

   * Den ASCII-Text bekommt man unter www.network-science.de/ascii/  * 

   ******************************************************************/ 

   MainPdat = "" !! , 

      "        __    __               .__               " !! , 

      "_____  |  | _|  | ____ _____  _|__| ____   ____  " !! , 

      "\__  \ |  |/ /  |/ /  |  \  \/ /  |/ ___\ / ___\ " !! , 

      " / __ \|    <|    <|  |  /\   /|  / /_/  > /_/  >" !! , 

      "(____  /__|_ \__|_ \____/  \_/ |__\___  /\___  / " !! , 

      "     \/     \/    \/             /_____//_____/  " !! , 

      "" 

   Shaddows = "éóúíöü" 

   "vput (MainPdat Shaddows) shared" 

   call akuv#col 

return 

 

close_tables: 

   /****************************************************************** 

   * Schließen der Kundentabelle                                     * 

   * Rückschreiben InPlace. Wenn nicht mögl. 50% Platz reservieren.  * 

   ******************************************************************/ 

   "tbclose &KdTAB replcopy pad(50)" 

   /****************************************************************** 

   * Schließen der Auftragstabelle                                   * 

   * Rückschreiben InPlace. Wenn nicht mögl. 50% Platz reservieren.  * 

http://www.network-science.de/ascii/

   ******************************************************************/ 

   "tbclose &AufTAB replcopy pad(50)" 

return 

 

end_tables: 

   /****************************************************************** 

   * Schließen der Tabellen ohne Speichern                           * 

   ******************************************************************/ 

   "tbend &KdTAB" 

   "tbend &AufTAB" 

return 

 

open_tables: 

   /****************************************************************** 

   * Öffnen der KD-Tab. ggfls TBCREATE, falls nicht vorhanden.       * 

   ******************************************************************/ 

   "tbopen &kdtab write" 

   select 

      when rc = 0 then nop 

      when rc = 8 then do 

         "tbcreate &KdTAB keys(&KdKeys) names(&KdNames) write" 

         if rc > 0 then call abend RC "RC:"rc "bei TBCREATE KDTAB" 

         "verase sortargk profile" 

      end 

      otherwise do 

         if rc = 12 then abendmsg = "RC:"rc "Kundentable in USE" 

         else            abendmsg = "RC:"rc "bei TBOPEN KDTAB" 

         call abend abendmsg 

      end 

   end 

   /****************************************************************** 

   * Öffnen der Auftrags-Tab. ggfls TBCREATE, falls nicht vorhanden. * 

   ******************************************************************/ 

   "tbopen &auftab write" 

   select 

      when rc = 0 then nop 

      when rc = 8 then do 

         "tbcreate &aufTAB names(&aufNames) write" 

         if rc > 0 then call abend RC "RC:"rc "bei TBCREATE AUFTAB" 

         "verase sortarga profile" 

      end 

      otherwise do 

         if rc = 12 then abendmsg = "RC:"rc "Auftragstabelle in USE" 

         else            abendmsg = "RC:"rc "bei TBOPEN AUFTAB" 

         call abend abendmsg 

      end 

   end 

return 

 

Set_Environment: 

   address tso "ALTLIB ACT APPL(EXEC) DATASET("RexxLib") UNCOND" 

   "libdef ispplib dataset id(&PanLib) stkadd" 

   "libdef ispmlib dataset id(&MsgLib) stkadd" 

   "libdef ispslib dataset id(&SklLib) stkadd" 

   "libdef ispTlib dataset id(&TblLib) stkadd" 

   "libdef ispTabl dataset id(&TblLib) stkadd" 

return 

 

Reset_Environment: 

   address tso "ALTLIB DEACT APPL(EXEC)" 

   "libdef ispplib" 

   "libdef ispmlib" 

   "libdef ispslib" 

   "libdef ispTlib" 

   "libdef ispTabl" 

return 

 

Check_Command_Table: 

   CmdTab = "AKUVCMDS" 

   TableDsn = left(RxLib,lastpos(".",RxLib))"TABLES" 

   "libdef isptlib dataset id('&TableDsn') stkadd" 

   if sysdsn("'"TableDsn"("CmdTab")'") = "OK" then return 

   /****************************************************************** 

   * Keine Command-Table gefunden - wir machen eine                  * 

   ******************************************************************/ 

   _=0 

   _=_+1;T._="RAUF~2~ALIAS UP~Alias für UP" 

   _=_+1;T._="RUNTER~2~ALIAS DOWN~Alias für DOWN" 

   _=_+1;T._="SMILY~3~SELECT CMD(%AKUV#FUN SMILY)~Lach doch mal" 

   _=_+1;T._="KAFFEE~3~SELECT CMD(%AKUV#FUN COFFEE)~Mach mal Pause" 

   _=_+1;T._="WELCOME~3~SELECT CMD(%AKUV#FUN WELCOME)~Guddn Dach" 

   _=_+1;T._="KILLROY~3~SELECT CMD(%AKUV#FUN KILLROY)~McKillroy was here" 

   "libdef isptabl dataset id('&TableDsn') stkadd" 

   "tbend &CmdTab" 

   "tbcreate &CmdTab names(ZCTVERB ZCTTRUNC ZCTACT ZCTDESC) write" 

   do §=1 to _ 

      parse var T.§ zctverb "~" zcttrunc "~" zctact "~" zctdesc 

      "tbadd &CmdTab" 

   end 

   "tbclose &CmdTab replcopy pad(50)" 

   "libdef isptabl" 

return 

 

abend: parse arg zedlmsg;address ispexec "setmsg msg(isrz000)";exit 

 

/* Programmbeschreibung ********************************************** 

 

   AKUVIGG ist das Einstiegtprogramm der Auftrags- und Kundenverwaltung 

   Im Wesentlichen erfüllt es folgende vorbereitende Aufgaben vor dem 

   Start des zur Anwendung gehörenden Selection-Panels: 

 

   ° Die globalen Variablen werden gesetzt und abhängig davon. ob 

     sie über LOGOFF hinaus benötigt werden, im ISPF-Profife- oder 

     Shared-Pool abgelegt. Um den Umgang mit diesen Variablen zu ver- 

     einfachen, werden die Namen aller Shared-Pool-Variablen in SHRVARS, 

     die der Profile-Pool-Variablen in PROFVARS abgelegt. 

     Auf diese Weise können in sämtlichen Unterprogrammen mit: 

     "vget (ShrVars) shared"      "vget (ProfVars)M profile" und 

     "vget (&Shrvars) shared"     "vget (&ProfVars) profile" 

     gelesen werden, ohne diese einzeln benennen zu müssen. 

 

   ° Um sicher zu stellen, dass die Profil-Pool-Variablen richtig zur 

     Verfügung stelhen, erfolgt der Start der Anwendung mit der 

     Applikationskennung AKUV. 

 

   ° Das Programm prüft eine Commandtable auf Existenz (AKUVCMD). 

     Wird diese nicht gefunden, wird sie im Programm erzeugt. 

 

   ° Die Umgebung (Dateilandschaft) für die Anwendung wird aufgebaut, 

     wobei die Namen der diversen Bibliotheken vom Namen der Startrexx- 

     Library abgeleitet wird. Bei Programmende wird diese Umgebung 

     wieder abgebaut. 

 

   ° Alle permanenten Tabellen werden zur Verfügung gestellt. Existiert 

     eine Tabelle nicht, wird sie entsprechend erzeugt. Dies hat den 

     Vorteil, dass sich alle Programme im weiteren Verlauf darauf 

     verlassen können, dass die permanenten Tabellen existieren. 

 

*/ 

4.17.8.2. AKUV#COL – Colorieren von AKUVIGG 

Es ist nur eine Spielerei, welche dafür sorgt, dass der AKUVIGG-Schriftzug bei Drücken der Enter-
taste im Selectionpanel in immer neuen Farben schillert. Was tut man nicht alles, wenn einem lang-

weilig ist     

/* REXX * Colorieren der Dynamic Area für AKUVIGG 

*/ 

address ispexec 

"vget (MainPdat Shaddows) shared" 

MainPshd = 

do §=1 to length(MainPdat) 

   MainPshd=MainPshd !! substr(Shaddows,random(1,length(Shaddows)),1) 

End 

"vput MainPshd shared" 

return 

4.17.8.3. AKUV#KUN – Das Kunden-Verwaltungsprogramm 

Das Kunden-Verwaltungsprogramm dient der Pflege der Kundentabelle, wobei es Querbezüge zu 
den Auftragsdaten gibt. Ein Kundensatz darf nicht gelöscht werden, wenn noch nicht verarbeitete 
Auftragssätze existieren (und dergleichen mehr). 

Hier werden verschiedene Dinge auf eine Art und Weise kodiert, welche zeigt, wie ein Komfortables 
Scannen oder Sortieren und dergleichen aussehen kann. Es geht natürlich auch immer ganz anders, 
was im AKUV#AUF, der Auftragsverwaltung, zu sehen ist. 

/* REXX * AKUF#KUN * Bearbeitung Kundentabelle 

*/ 

address ispexec 

"control errors cancel" 

 

main_start: 

   call set_vars         /* Setzen der Variablen            */ 

   call table_show       /* Anzeige und Bearbeitung KdTAB   */ 

main_ende: 

 

exit 

 

/********************************************************************* 

* Subroutinen                                                        * 

*********************************************************************/ 

 

AuftragDA: 

   /****************************************************************** 

   * Prüft offene Aufträge                                           * 

   * Wird kein Parameter übergeben, wird nur geprüft, ob mind. EIN   * 

   * Auftrag vorhanden ist und Löschen des Kundensatzes verhindert   * 

   * Mit Parm-Übergabe werden die vorh. Auftragssätze gezählt.       * 

   ******************************************************************/ 

   aufCNT = 0 

   "tbvclear &auftab" 

   $aufKDNR = #KDNR 

   "tbsarg &auftab" 

   "tbscan &auftab" 

   do while rc = 0 

      if $aufVKZ <> "ERL" then do 

         zedlmsg = "Offene Aufträge zu KdNr" #KDNR "vorhanden" 

         aufCNT = aufCNT + 1 

         if arg(1) = "" then leave 

      end 

      "tbscan &auftab" 

   End 

   "setmsg msg(isrz000)" 

return (aufCNT > 0) 

 

DatenOK: 

   /****************************************************************** 

   * Plausibilitätsprüfungen, die im Datenpanel nicht so ohne        * 

   * weiteres möglich sind                                           * 

   ******************************************************************/ 

   if #gebdat = "" then return 1 

   parse var #gebdat 1 tt 3 f1 4 mm 6 f2 7 jjjj 

   mtab = "31" 28+((jjjj//4=0)-(jjjj//100=0)+(jjjj//400=0)) , 

          "31 30 31 30 31 31 30 31 30 31" 

   Select 

      when mm < 1 ! mm > 12            then msg = "Monat falsch" 

      when tt < 1 ! tt > word(mtab,mm) then msg = "Tag falsch" 

      when jjjj < word(date(),3)-100   then msg = "Jahr zu klein" 

      when jjjj!!mm!!tt > date("S")    then msg = "Datum zu groß" 

      otherwise msg = 

   end 

   if msg <> "" then do 

      zedlmsg = msg 

      "setmsg msg(isrz000)" 

      csrfld = "#gebdat" 

   end 

return (msg = "") 



Inside z/OS Seite 294 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

 

LCMD_Auftrag: 

   call AuftragDA "?" 

   zedlmsg = aufCNT "offene Aufträge zu Kundennummer" #KDNR 

   "setmsg msg(isrz000)" 

Return 

 

LCMD_Browse: 

   do until PanRC > 0 

      PanRC = Show_Kunden_Satz() 

   End 

Return 

 

LCMD_Delete: 

   if Show_Kunden_Satz() > 0 then return 

   if AuftragDA()            then return 

   "tbdelete &kdtab" 

Return 

 

LCMD_Insert: 

   if Arg() > 0 then LCmd = Arg(1) 

   "tbvclear &kdtab" 

   do forever 

      #KDNR = LstKdnr +1 

      if Show_Kunden_Satz() > 0 then leave 

      if ^DatenOK() then iterate 

      call trans_2_upper 

      "tbadd &kdtab order" 

      LstKdnr = #KDNR 

      "vput LstKdnr profile" 

      "tbvclear &kdtab" 

   End 

Return 

 

LCMD_Update: 

   do forever 

      PanRC = Show_Kunden_Satz() 

      if PanRC > 0 then leave 

      if ^DatenOK() then iterate 

      call trans_2_upper 

      "tbmod &kdtab order" 

      Leave 

   End 

Return 

 

Show_Kunden_Satz: 

   zWinTTL = WinTtl.lcmd 

   PanMsg  = PanMsg.lcmd 

   "addpop row(5) column(12)" 

   "display panel(AKUV#KDD)" 

   DispRC = RC 

   "rempop" 

return DispRC 

Trans_2_upper: 

   #NNameU = translate(#NName) 

   #VNameU = translate(#VName) 

   #OrtU   = translate(#Ort  ) 

   #StrU   = translate(#Str  ) 

return DispRC 

 

PCMD_Geheim: 

   /****************************************************************** 

   * Interpretative Command-Ausführung für Administratoren           * 

   ******************************************************************/ 

   if ^Admin then do 

      zedlmsg = "DU nicht, Schnarchnase." 

      "setmsg msg(isrz000)" 

      Return 

   End 

   do forever 

      say address() ">> Bliehs enda jur Wischis>" 

      parse external cmd 

      if cmd = "" then leave 

      interpret cmd 

   end 

return 

 

PCMD_Reset_KdTab_Filter: 

   "tbvclear &kdtab" 

   parse value "*" with , 

      1 #nnameu  1 #vnameu 1 #plz 1 #ortu 1 stru , 

      1 Snname   1 Svname  1 Splz 1 Sort  1 Sstr 

   "tbsarg &kdtab" 

   crpKD    = 1 

return 

 

PCMD_Sort: 

   arg SortArgK 

   if wordpos(SortArgK,"K N V P O S") = 0 then do 

      zedlmsg = "Sortparameter ist Stelle 1 der Spaltenüberschrift." 

      "setmsg msg(isrz000)" 

      Return 

   End 

   call set_colors_4_table_cols "TURQ GREEN" 

   "tbsort &kdtab fields("sort.SortArgK")" 

   "vput sortargk profile" 

   crpKD    = 1 

return 

 

set_kdtab_filter: 

   /****************************************************************** 

   * Filtern der Kundentabelle                                       * 

   ******************************************************************/ 

   crpKD    = 1 

   "tbvclear &kdtab" 

   #nnameu  = translate(Snname) 

   #vnameu  = translate(Svname) 

   #plz     = Splz 

   #ortu    = translate(Sort) 

   #stru    = translate(Sstr) 

   "tbsarg &kdtab" 

Return 

 

set_vars: 

   /****************************************************************** 

   * Initialisierung der Variablen.                                  * 

   ******************************************************************/ 

   KdShow   = "SCAN" 

 

   sort.k   = "#kdnr,n,a" 

   sort.n   = "#nnameu,c,a,#vnameu,c,a,#ortu,c,a,#plz,n,a" 

   sort.v   = "#vnameu,c,a,#nnameu,c,a,#ortu,c,a,#plz,n,a" 

   sort.p   = "#plz,n,a,#ortu,c,a,#stru,c,a" 

   sort.o   = "#ortu,c,a,#plz,n,a,#stru,c,a" 

   sort.s   = "#stru,c,a,#nnameu,c,a,#vnameu,c,a" 

 

   "vget (shrvars)  shared" 

   "vget (&shrvars)  shared" 

   "vget (profvars) profile" 

   "vget (&profvars) profile" 

   if SortArgK = "" then call PCMD_Sort "N" 

 

   PrimCMDS = "I RES SORT KAL" userid() 

   LineCMDS = "A B D I U" 

   LCmdOK   = translate(LineCMDS,","," ") 

 

   PCMDS    = "}I{(Insert Satz)}SORT{parm}RES{(Reset Filter)" !! , 

              "}KAL{(Kalender)" 

   LCMDS    = "}A{(Aufträge?)" !! , 

              "}B{(Browse)}D{(Delete)}I{(Insert)}U{(Update)" 

   kdLCMDS  = "A,B,D,I,U" 

 

   Snname   = "*" 

   Svname   = "*" 

   Splz     = "*" 

   Sort     = "*" 

   Sstr     = "*" 

   call set_colors_4_table_cols "TURQ GREEN" 

 

   WinTtl.B    = "Anzeige Kundensatz" 

   PanMsg.B    = "}END{(Rückkehr zur Kundentabelle)" 

   WinTtl.D    = "Löschen Kundensatz" 

   PanMsg.D    = "}ENTER{(Satz löschen)}END{(Abbruch)" 

   WinTtl.I    = "Erfassen Kundensatz" 

   PanMsg.I    = "}ENTER{(Neuanlage) }END{(Abbruch)" 

   WinTtl.U    = "Ändern Kundensatz" 

   PanMsg.U    = "}ENTER{(Ändern) }END{(Abbruch)" 

Return 

 

set_colors_4_table_cols: 

   arg NotSorted Sorted 

   parse value NotSorted with , 

      1 farbe1 1 farbe2 1 farbe3 1 farbe4 1 farbe5 1 farbe6 

 

   interpret "farbe"pos(SortArgK,"KNVPOS") "=" Sorted 

return 

 

table_show: 

   /****************************************************************** 

   * Anzeige und Bearbeitung der Kundentabelle                       * 

   ******************************************************************/ 

   call pcmd_reset_KdTab_filter 

   crpKD = 1 

   do forever 

      if doTRACE = "J" then trace ?r 

      call set_kdtab_filter 

      "tbtop &kdtab" 

      "tbskip &kdtab number(&CRPKD)" 

      "tbdispl &kdtab panel(AKUV#KDT)" 

      if RC >= 8 then leave 

      crpKD = ZTDTOP 

      drop zedsmsg zedlmsg   /* falls vom Vorlauf noch besetzt */ 

      parse upper var zcmd pcmd parms 

      if wordpos(PCMD,PrimCMDS) = 0 & PCMD <> "" then do 

         zedlmsg = "Der Befehl" PCMD "ist nicht erlaubt" 

         "setmsg msg(isrz000)" 

         ztdsels = 0 

         iterate 

      end 

      select 

         when PCMD = userid() then call PCMD_Geheim 

         when PCMD = "I"      then call LCMD_Insert "I" 

         when PCMD = "KAL"    then call AKUV#KAL parms 

         when PCMD = "RES"    then call PCMD_Reset_KdTab_Filter 

         when PCMD = "SORT"   then call PCMD_Sort left(parms,1) 

         otherwise nop 

      end 

      do while ZTDSELS > 0 

         "control display save" 

         Select 

            when LCMD = "A"   then call LCMD_Auftrag 

            when LCMD = "B"   then call LCMD_Browse 

            when LCMD = "D"   then call LCMD_Delete 

            when LCMD = "I"   then call LCMD_Insert 

            when LCMD = "U"   then call LCMD_Update 

            otherwise nop 

         end 

         "control display restore" 

         if ZTDSELS > 1 then "tbdispl &kdtab" 

         else ZTDSELS = 0 

      end 

   end 

return 

 

abend: 

   /****************************************************************** 

   * ABEND-Routine.                                                  * 

   ******************************************************************/ 

   arg RetCode AbendMSG 

   if sysvar(sysenv) = "FORE" then do 

      zedlmsg = AbendMSG 

      "setmsg msg(isrz000)" 

   End 

   else do 

      say AbendMSG 

   end 

exit RetCode 

4.17.8.4. AKUV#AUF – Die Auftragsverwaltung 

Um möglichste viele unterschiedliche Beispiele für eine Lösung aufzuzeigen, ist in der Auftragsver-
waltung das Eine oder andere auf andere Art als in der Kundenverwaltung kodiert. 

/* REXX * AKUV#AUF * Bearbeitung Auftragstabelle 

*/ 

address ispexec 

"control errors return" 

 

main_start: 

   call set_vars 

   call table_show 

main_ende: 

 

exit 

 

Kunde_unbekannt: 

   /****************************************************************** 

   * Prüft Kundennummer in Kundentabelle auf vorhanden               * 

   ******************************************************************/ 

   arg #kdnr 

   "tbtop &kdtab" 

   "tbscan &kdtab arglist(#kdnr) condlist(eq)" 

return (rc <> 0) 

 

LCMD_D: 

   /****************************************************************** 

   * Löschen Zeile in Auftragstabelle                                * 

   ******************************************************************/ 

   "tbdelete &auftab" 

Return 

 

LCMD_I: 

LCMD_U: 

LCMD_G: 

   /****************************************************************** 

   * _I: Insert Satz in Auftragstabelle                              * 

   * _U: Update Satz in Auftragstabelle                              * 

   * _G: Update Satz in Auftragstabelle (erweitertes Datenpanel)     * 

   ******************************************************************/ 

   if arg() > 0 then lcmd = arg(1) 

   zwinttl = winttl.lcmd 

   panmsg1 = pmsg1.lcmd 

   panmsg2 = pmsg2.lcmd 

   if lcmd = "I" then "tbvclear &auftab" 

   "addpop row(5) column(12)" 

   do forever 

      if lcmd = "G" then "display panel(AKUV#AUG)" 

      else "display panel(AKUV#AUD)" 

      if RC > 0 then leave 

      if Kunde_unbekannt($AufKDNR) then do 

         zedlmsg = "KundenNr" $aufKDNR "in KD-Tabelle nicht vorhanden" 

         "setmsg msg(isrz000)" 

         Iterate 

      End 

      $artEPR = format($artEPR,,2) 

      if wordpos(lcmd,"G U") > 0 then do 

         "tbput &auftab order" 

         Leave 

      End 

      $aufDAT = date("S") 

      "tbadd &auftab order" 

      "tbvclear &auftab" 

   End 

   "rempop" 

Return 

 

PCMD_Reorg_Auftab: 

   /****************************************************************** 

   * Entfernen Auftragssätze mit $aufVKZ = "ERL"                     * 

   ******************************************************************/ 

   cntDEL = 0 

   "tbtop &auftab" 

   $aufVKZ = "ERL" 

   "tbscan &auftab arglist($aufVKZ) condlist(EQ)" 

   do while rc = 0 

      "tbdelete &auftab" 

      cntDEL = cntDEL + 1 

      "tbscan &auftab arglist($aufVKZ) condlist(EQ)" 

   End 

   if cntDEL = 0 then do 

      zedlmsg = "Keine erledigten Auftragssätze gefunden." 

   End 

   else do 

      if cntDEL = 1 then zedlmsg = "1 erledigten Auftragssatz gelöscht." 

      else zedlmsg = cntDEL "erledigte Auftragssätze gelöscht." 

   End 

   "setmsg msg(isrz000)" 

   crpAUF = 1 

return 

 

PCMD_Scan: 

   /****************************************************************** 

   * Scan auf Kundennummer oder Gesamtanzeige                        * 

   ******************************************************************/ 

   Parse Arg feld wert 

   upper feld 

   "tbvclear &auftab" 

   AufShow = "SCAN" 

   Select 

      when feld = "KDN" then $AufKDNR = wert 

      when feld = "BEZ" then $ArtBez  = wert 

      when feld = "DAT" then $AufDAT  = wert 

      when feld = "VKZ" then $AufVKZ  = translate(wert) 

      otherwise AufShow = "ALL" 

   end 

   "tbsarg &auftab" 

   crpAUF = 1 

return 

 

PCMD_Sort: 

   /****************************************************************** 

   * Sortierung der Auftragstabelle                                  * 

   ******************************************************************/ 

   arg SortArgA 

   if wordpos(SortArgA,"ANZ BEZ DAT KDN NUM VKZ") = 0 then do 

      zedlmsg = "Mögliche Sortparameter sind: BEZ, DAT, KDN, NUM, VKZ" 

      "setmsg msg(isrz000)" 

      Return 

   End 

   "vput SortArgA profile" 

   sflds = sort.SortArgA 

   "tbsort &aufTAB fields(&sflds)" 

   sortiert = feld.SortArgA 

   crpAUF = 1 

return 

 

set_model: 

   /****************************************************************** 

   * Definition der aktuellen Ansicht.                               * 

   ******************************************************************/ 

   arg asicht 

   header = header.asicht 

   modset = modset.asicht 

   zfield = zfield.asicht 

   lineCM = lnCMDs.asicht 

   "vput (asicht) profile" 

Return 

 

set_vars: 

   /****************************************************************** 

   * Initialisierung der Variablen.                                  * 

   ******************************************************************/ 

   "vget (ShrVars)   shared" 

   "vget (&ShrVars)  shared" 

   "vget (ProfVars)  profile" 

   "vget (&ProfVars) profile" 

 

   aufLCMDS = "D,I,U" 

   if Admin then aufLCMDS = aufLCMDS",G" 

 

   AufShow  = "ALL" 

 

   sort.bez = "$ArtBEZ,c,a" 

   sort.dat = "$aufDAT,N,A,$AufKDNR,N,A" 

   sort.kdn = "$aufKDNR,N,A,$AufDAT,N,D" 

   sort.num = "$ArtNUM,N,A" 

   sort.anz = "$ArtANZ,N,A" 

   sort.vkz = "$aufVKZ,C,A,$AufDATS,N,D" 

   feld.bez = "Artikel-Bezeichnung" 

   feld.dat = "Auftragsdatum" 

   feld.kdn = "Kundennummer" 

   feld.num = "Artikelnummer" 

   feld.anz = "Artikelanzahl" 

   feld.vkz = "Auftrags-Verarbeitungskennzeichen" 

 

   "vget (asicht SortArgA) profile" 

   if Asicht = "" then Asicht = 1 

   if SortArgA = "" then do 

      call pcmd_sort "KDN" 

   end 

   else sortiert = feld.SortArgA 

 

   winttl.G = "General Update Administrator" 

   pmsg1.G  = "}Alle Eingabefelder sind MUSS-Felder" 

   pmsg2.G  = "}ENTER{(Ändern)}END{(Abbruch)" 

 

   winttl.I = "Erfassen Auftragssatz" 

   pmsg1.I  = "}Alle Eingabefelder sind MUSS-Felder" 

   pmsg2.I  = "}ENTER{(Erfassen)}END{(Abbruch)" 

 

   winttl.U = "Ändern Auftragssatz" 

   pmsg1.U  = "}Alle Eingabefelder sind MUSS-Felder" 

   pmsg2.U  = "}ENTER{(Ändern)}END{(Abbruch)" 

 

   header.1 = "³V"                    !! ,  /* LCMD-SpaltenKZ     */ 

              "²"left("AufKDN",6)     !! ,  /* Auftrags-Kundennr. */ 

              "²"left("ArtNUM",6)     !! ,  /* Artikelnummer      */ 

              "²"left("ArtANZ",6)     !! ,  /* Atrikel Anzahl     */ 

              "²"left("ArtBEZ",33)    !! ,  /* Artikelbezeichnung */ 

              "²"left("E-Preis",7)    !! ,  /* Einzelpreis        */ 

              "²"left("VKZ",3)        !! ,  /* VerarbeitungsKZ    */ 

              "²"left("AuftrDAT",8)   !! ,  /* Auftragsdatum      */ 

              "³" 

 

   modset.1 = "~"left("Z",1)          !! ,  /* LCMD (INPUT)       */ 

              "µ"left("Z",6)          !! ,  /* $AufKDNR           */ 

              "µ"left("Z",6)          !! ,  /* $ArtNUM            */ 

              "µ"left("Z",6)          !! ,  /* $ArtANZ            */ 

              "\"left("Z",33)         !! ,  /* $ArtBEZ            */ 

              "µ"left("Z",7)          !! ,  /* $ArtEPR            */ 

              "µ"left("Z",3)          !! ,  /* $AufVKZ            */ 

              "µ"left("Z",8)          !! ,  /* $AufDAT            */ 

              "+" 

 

   zfield.1 = "LCMD"                     ,  /* Line Command       */ 

              "$AufKDNR"                 ,  /* Auftrags-Kundennr. */ 

              "$ArtNUM"                  ,  /* Artikelnummer      */ 

              "$ArtANZ"                  ,  /* Atrikel Anzahl     */ 

              "$ArtBEZ"                  ,  /* Artikelbezeichnung */ 

              "$ArtEPR"                  ,  /* Einzelpreis        */ 

              "$AufVKZ"                  ,  /* VerarbeitungsKZ    */ 

              "$AufDAT"                  ,  /* Auftragsdatum      */ 

              !! "" 

 

   lnCMDS.1 = "³D^(Delete) ³I^(Insert) ³U^(Update)" 

 

   header.2 = "³V"                    !! ,  /* LCMD-SpaltenKZ     */ 

              "²"left("ArtNUM",6)     !! ,  /* Artikelnummer      */ 

              "²"left("ArtANZ",6)     !! ,  /* Atrikel Anzahl     */ 

              "²"left("ArtBEZ",33)    !! ,  /* Artikelbezeichnung */ 

              "²"left("E-Preis",7)    !! ,  /* Einzelpreis        */ 

              "²"left("VKZ",3)        !! ,  /* VerarbeitungsKZ    */ 

              "²"left("AuftrDAT",8)   !! ,  /* Auftragsdatum      */ 

              "²"left("AufKDN",6)     !! ,  /* Auftrags-Kundennr. */ 

              "³" 

 

   modset.2 = "~"left("Z",1)          !! ,  /* LCMD (INPUT)       */ 

              "µ"left("Z",6)          !! ,  /* $ArtNUM            */ 

              "µ"left("Z",6)          !! ,  /* $ArtANZ            */ 

              "\"left("Z",33)         !! ,  /* $ArtBEZ            */ 

              "µ"left("Z",7)          !! ,  /* $ArtEPR            */ 

              "µ"left("Z",3)          !! ,  /* $AufVKZ            */ 

              "µ"left("Z",8)          !! ,  /* $AufDAT            */ 

              "µ"left("Z",6)          !! ,  /* $AufKDNR           */ 

              "+" 

 

   zfield.2 = "LCMD"                     ,  /* Line Command       */ 

              "$ArtNUM"                  ,  /* Artikelnummer      */ 

              "$ArtANZ"                  ,  /* Atrikel Anzahl     */ 

              "$ArtBEZ"                  ,  /* Artikelbezeichnung */ 

              "$ArtEPR"                  ,  /* Einzelpreis        */ 

              "$AufVKZ"                  ,  /* VerarbeitungsKZ    */ 

              "$AufDAT"                  ,  /* Auftragsdatum      */ 

              "$AufKDNR"                 ,  /* Auftrags-Kundennr. */ 

              !! "" 

 

   lnCMDS.2 = "³D^(Delete) ³I^(Insert) ³U^(Update)" 

 

   header.3 = "³V"                    !! ,  /* LCMD-SpaltenKZ     */ 

              "²"left("AufKDN",6)     !! ,  /* Auftrags-Kundennr. */ 

              "²"left("ArtNUM",6)     !! ,  /* Artikelnummer      */ 

              "²"left("ArtANZ",6)     !! ,  /* Atrikel Anzahl     */ 

              "²"left("ArtBEZ",33)    !! ,  /* Artikelbezeichnung */ 

              "²"left("E-Preis",7)    !! ,  /* Einzelpreis        */ 

              "²"left("VKZ",3)        !! ,  /* VerarbeitungsKZ    */ 

              "²"left("AuftrDAT",8)   !! ,  /* Auftragsdatum      */ 

              "³" 

 

   modset.3 = "\"left("Z",1)          !! ,  /* LCMD (INPUT)       */ 

              "µ"left("Z",6)          !! ,  /* $AufKDNR           */ 

              "µ"left("Z",6)          !! ,  /* $ArtNUM            */ 

              "µ"left("Z",6)          !! ,  /* $ArtANZ            */ 

              "\"left("Z",33)         !! ,  /* $ArtBEZ            */ 

              "µ"left("Z",7)          !! ,  /* $ArtEPR            */ 

              "µ"left("Z",3)          !! ,  /* $AufVKZ            */ 

              "µ"left("Z",8)          !! ,  /* $AufDAT            */ 

              "+" 

 

   zfield.3 = "LCMD"                     ,  /* Line Command       */ 

              "$AufKDNR"                 ,  /* Auftrags-Kundennr. */ 

              "$ArtNUM"                  ,  /* Artikelnummer      */ 

              "$ArtANZ"                  ,  /* Atrikel Anzahl     */ 

              "$ArtBEZ"                  ,  /* Artikelbezeichnung */ 

              "$ArtEPR"                  ,  /* Einzelpreis        */ 

              "$AufVKZ"                  ,  /* VerarbeitungsKZ    */ 

              "$AufDAT"                  ,  /* Auftragsdatum      */ 

              !! "" 

 

   lnCMDS.3 = "³keine" 

return 

 

table_show: 

   /****************************************************************** 

   * Anzeige und Bearbeitung der Auftragstabelle                     * 

   ******************************************************************/ 

   crpAUF = 1 

   do forever 

      header = header.asicht 

      modset = modset.asicht 

      zfield = zfield.asicht 

      lineCM = lnCMDs.asicht 

      "tbtop &auftab" 

      "tbskip &auftab number(&CRPAUF)" 

      "tbdispl &auftab panel(AKUV#AUT)" 

      if RC >= 8 then leave 

      crpAUF = ZTDTOP 

      drop zedsmsg zedlmsg   /* falls vom Vorlauf noch besetzt */ 

      parse var zcmd pcmd parms 

      upper PCMD 

      select 

         when PCMD = "I"      then call LCMD_I "I" 

         when PCMD = "R"      then call PCMD_Reorg_Auftab 

         when PCMD = "RES"    then call PCMD_Scan 

         when PCMD = "SCAN"   then call PCMD_Scan parms 

         when PCMD = "SORT"   then call PCMD_Sort parms 

         when wordpos(PCMD,"1 2 3") > 0 then call set_model pcmd 

         otherwise nop 

      end 

      do while ZTDSELS > 0 

         "control display save" 

         interpret "CALL LCMD_"lcmd 

         "control display restore" 

         if ZTDSELS > 1 then "tbdispl &auftab" 

         else ZTDSELS = 0 

      end 

   end 

return 

 

abend: 

   /****************************************************************** 

   * ABEND-Routine.                                                  * 

   ******************************************************************/ 

   arg RetCode ZEdLMsg 

   if sysvar(sysenv) = "FORE" then "setmsg msg(isrz000)" 

   else say AbendMSG 

exit RetCode 

 

/* Programmbeschreibung ********************************************** 

   Bearbeitung der Auftragstabelle 

*/ 

4.17.8.5. AKZV#SRE – Auswahl für das Rechnungsprogramm 

/* REXX * AKUV#SRE * Auswahl Einzel-/Monatsrechnung 

*/ 

address ispexec 

"control errors return" 

 

"addpop row(9) column(12)" 

"select panel(AKUV#SRE)" 

"rempop" 

Exit 

4.17.8.6. AKUV#RE – Die Rechnungserstellung 

Das Programm ist so geschrieben, dass es anhand des Formats des Übergabeparameters erkennen 
kann, ob eine Kunden-Einzelrechnung oder ein Gesamt-Rechnunglauf für einen bestimmten Monat 
durchführen soll. Bei Einzelrechnung wird die Kundennummer übergeben. Für die Monatsrechnung 
wird eine Wortkette mit Kundennummern gebildet, welche für den betreffenden Monat offene Posten 
haben. 

/* REXX * AKUV#RE * Rechnungsprogramm 

*/                                                                          

arg PARM . 

 

call msg "OFF" 

 

address ispexec 

"control errors return" 

"tbtop &auftab" 

"tbtop &kdtab" 

 

call set_vars 

if length(PARM) = 6 then kdnr = get_kdnr() 

else                     kdnr = kundeDA(parm) 

do $ = 1 to words(kdnr) 

   kunde = word(kdnr,$) 

   call rechnung kunde 

end 

 

exit 

 

rechnung: 

   _C = 0 

   drop out. 

   _snetto = 0 

   "tbvclear &kdtab" 

   "tbtop &kdtab" 

   #kdnr = arg(1) 

   "tbsarg &kdtab" 

   "tbscan &kdtab" 

   call build_rechnungskopf 

   "tbvclear &auftab" 

   "tbtop &auftab" 

   $aufkdnr = #kdnr 

   "tbsarg &auftab" 

   "tbscan &auftab" 

   do while rc = 0 

      call build_rechnung 

      $aufVKZ = "ERL" 

      "tbput &auftab" 

      "tbscan &auftab" 

   End 

   call build_rechnungsende 

   call write_rechnung 

   "vput LstRechN profile" 

   "tbsave &auftab" 

Return 

 

build_rechnungskopf: 

   rabatt = word("0 5 10 20 25",wordpos(#KDGR,"0 1 2 3 4")) 

   _C=_C+1;out._C = absender 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = Strip(#vname) strip(#nname) 

   _C=_C+1;out._C = Strip(#str) 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = Strip(#plz) #ort 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = " " 

   LstRechN = LstRechN + 1 

   _C=_C+1;out._C = "Rechnung" LstRechN "vom" Heute", KdNr:" #kdnr 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = ueber_1 

   _C=_C+1;out._C = ueber_2 

Return 

 

build_rechnung: 

   netto = $artanz * $artEPR 

   _snetto = _snetto + netto 

   _C=_C+1;out._C = "" !!            ,  /* Zeilenaufbau     */ 

        §datd($aufdat)               ,  /* Auftragsdatum    */ 

        right($artnum,6)             ,  /* Artikelnummer    */ 

        right(§fmt($artanz),7)       ,  /* Artikelanzahl    */ 

        left($artbez,25)             ,  /* Art.-Bezeichnung */ 

        right(§fmt($artEPR),11)      ,  /* Einzelpreis      */ 

        right(§fmt(netto),14)        ,  /* Gesamtpreis      */ 

        "" 

Return 

 

build_rechnungsende: 

   # = 63 

   _C=_C+1;out._C = ueber_2 

   _C=_C+1;out._C = right("Gesamt",#) right(§fmt(_snetto),14) 

   _C=_C+1;out._C = right(" ",#)             right("-",14,"-") 

   if rabatt > 0 then do 

      _nachlass = (_snetto / 100) * rabatt 

      _snetto= _snetto - _nachlass 

      _C=_C+1;out._C = right("-" rabatt"% Rabatt",#) , 

                             right(§fmt(_nachlass),14) 

      _C=_C+1;out._C = right(" ",#)            right("-",14,"-") 

      _C=_C+1;out._C = right("Netto",#) right(§fmt(_snetto),14) 

      _C=_C+1;out._C = right(" ",#)            right("-",14,"-") 

   End 

   _mwst = format(_snetto * (mwst / 100),,2) 

   _endp = format(_snetto + _mwst,,2) 

   _C=_C+1;out._C = right("+"mwst"% MWSt",#) right(§fmt(_mwst),14) 

   _C=_C+1;out._C = right(" ",#) right("-",14,"-") 

   _C=_C+1;out._C = right("Endpreis",#) right(§fmt(_endp),14) 

   _C=_C+1;out._C = right(" ",#) right("=",14,"=") 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = "Zahlung bis spätestens" zahlziel() 

   _C=_C+1;out._C = "auf eines der nachfolgenden Konten" 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = BkHeader 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = BkVerb1 

   _C=_C+1;out._C = BkVerb2 

   _C=_C+1;out._C = " " 

   _C=_C+1;out._C = "Sonst kracht's" 

Return 

 

write_rechnung: 

   address tso 

   q1 = "RE"right(LstRechN,6,0) 

   q2 = "KD"right(#kdnr,6,0) 

   q3 = "D"left(date("S"),4) !! right(date("J"),3) 

   dsnRE = userid()"."q1"."q2"."q3 

   "del '"dsnRE"'" 

   "alloc dd(rech) da('"dsnRE"') new catalog space(1 1) tracks" , 

      "lrecl(80) blksize(0) recfm(F B) dsorg(PS)" 

   "execio * diskw rech (stem out. finis" 

   "free dd(rech)" 

   address ispexec 

   "view dataset('&dsnRE')" 

Return 

 

set_vars: 

   "vget (ShrVars)   shared" 

   "vget (&ShrVars)  shared" 

   "vget (ProfVars)  profile" 

   "vget (&ProfVars) profile" 

   ueber_1  = "Auftr-Dat. ArtNum Anzahl " , 

              "Bezeichnung               Einzelpreis    Gesamtpreis" 

   ueber_2  = "---------- ------ -------" , 

              "------------------------- ----------- --------------" 

Return 

 

get_kdnr: procedure 

   numbers = 

   "tbtop &auftab" 

   "tbskip &auftab" 

   do while rc = 0 

      if $aufvkz <> "ERL" & wordpos($aufkdnr,numbers) = 0 then do 

         numbers = strip(numbers $aufkdnr) 

      end 

      "tbskip &auftab" 

   End 

return numbers 

 

kundeDA: procedure 

   "tbtop &kdtab" 

   "tbvclear &kdtab" 

   #kdnr = arg(1) 

   "tbsarg &kdtab" 

   "tbscan &kdtab" 

   if rc <> 0 then call abend "Kundennummer" #kdnr "nicht in Kd-Tab." 

return #kdnr 

 

§fmt: procedure 

   parse value arg(1) with  INT "." DEC 

   Druck = reverse(strip(space( , 

           translate("abc.def.ghi.jkl.mno.pqr.stu.vwx.yz", , 

           reverse(int),"abcdefghijklmnopqrstuvwxyz") , 

           ,0),,".")) 

   if DEC <> "" then Druck = Druck","DEC 

return Druck 

 

§datd:    return translate("Tt.Mm.CcJj",arg(1),"CcJjMmTt") 

zahlziel: return §datd(date("S",date("B")+Ziel,"B")) 

abend:    parse arg zedlmsg;"setmsg msg(isrz000)";exit 12 

4.17.8.7. AKUV#DMY – Dummy für Info während der Entwicklung 

Während der Entwicklung eines Dialogs werden nie alle Programme auf einmal geschrieben (warum 
eigentlich?). Um Abbrüche wegen fehlender Programme zu vermeiden, wird für das noch zu Erstel-
lende einfach ein Dummy eingesetzt, der anstelle eines Abbruchs eine entsprechende Meldung aus-
gibt. Zu sehen im Auswahlpanel bei Eingabe von <23>. 

/* REXX 

*/ 

zedlmsg = arg(1) "is under Construction - try later" 

address ispexec "setmsg msg(isrz001)" 

exit 

4.17.8.8. AKUV#ALI – Kundenliste alphabetisch 

Alle wichtigen Hinweise sind in der Programmbeschreibung am Ende aufgeführt. 

/* REXX * AKUV#ALI * Erstellen Adressliste 

   Beschreibung am Programmende 

*/ 

 

lpp = 64 /* Seitenwechsel bei Druckausgabe */ 

 

parse source . . RxName . 

address ispexec 

"control errors return" 

"vget (ShrVars) shared" 

"vget (ztempn &ShrVars) shared" 

 

"tbquery &kdtab sortflds(AltSort)" 

AltSort = strip(translate(AltSort,"","()")) 

"tbsort &kdtab fields(#NNAMEU,C,A,#VNAMEU,C,A)" 

 

"ftopen temp" 

SkelINFO = "Job-Anweisung für Adressliste Kunde" 

"ftincl AKUV#JOB" 

SkelINFO = "IEBGENER für Adressliste Kunde" 

"ftincl AKUV#GE1" 

"ftincl AKUV#ADT" 

"ftclose" 

 

"tbsort &kdtab fields(&AltSort)" 

 

if Admin then do 

   zedlmsg = "Prüfe, ob alles OK ist" 

   "setmsg msg(isrz000)" 

   call #ediv "/V" /* User Defined Function – im ISPF-Teil beschrieben */ 

end 

else address tso "submit '"ztempf"'" 

 

exit 

/* Programmbeschreibung *********************************************** 

 

   Die Verarbeitung erfolgt über FileTailoring. 

 

   Da die Kundentabelle zur Laufzeit unter Umständen nicht so sortiert 

   ist, wie sie für die Alphaliste gebraucht wird, ermittelt die 

   Verarbeitung die aktuelle Sorierung, sortiert um und stellt am Ende 

   die ursprüngliche Sortierung wieder her. 

 

   Skel AKUB#JOB liefert die Job-Anweisung 

   Skel AKUB#GE1 liefert die JCL für den IEBGENER 

                 ist die Variable DSNUT1 gefüllt, werden die Daten von 

                 dieser Datei gelesen, ansonsten über INSTREAM 

   Skel AKUB#ADT holt die Daten der Adressliste aus der Kundentabelle 

*/ 

4.17.8.9. AKUV#TL1 – Telefonliste Variante 1 

Die eigentliche Verarbeitung erfolgt im Batch. Die dazu nötige JCL wird im Stack abgelegt und Schal-
ter-abhängig entweder im View angezeigt oder an den JES Internal Reader übergeben. 

/* REXX * AKUV#TL1 * Erstellen Telefonliste 

*/ 

test = 1 

test = 0 

§ = 0 

name. = left(" ",35) 

address ispexec 

"control errors return" 

"vget (ShrVars) shared" 

"vget (ztempn &ShrVars) shared" 

"tbquery &kdtab sortflds(AltSort)" 

AltSort = strip(translate(AltSort,"","()")) 

"tbsort &kdtab fields(#NNAMEU,C,A,#VNAMEU,C,A)" 

do forever 

   "tbskip &kdtab" 

   if rc > 0 then leave 

   if space(#TELFN #TELMOB #FAX,0) = "" then iterate 

   $=0 

   if strip(#telfn) <> "" then do 

      $=$+1;tel.$ = "Festnetz:" #telfn 

   End 

   if strip(#telmob) <> "" then do 

      $=$+1;tel.$ = "Mobil:   " #telmob 

   End 

   if strip(#fax) <> "" then do 

      $=$+1;tel.$ = "Telefax: " #fax 

   End 

   name.1 = left(strip(#nname)"," strip(#vname),35,".") 

   do _ = 1 to $ 

      §=§+1;out.§ = name._ tel._ 

   End 

End 

"tbsort &kdtab fields(&AltSort)" 

address tso 

"newstack" 

queue "//TELVERZ EXEC PGM=IEBGENER" 

queue "//SYSIN    DD DUMMY" 

queue "//SYSPRINT DD DUMMY" 

queue "//SYSUT2   DD SYSOUT=*" 

queue "//SYSUT1   DD *" 

do §=1 while out.§ <> "OUT."§ 

   queue out.§ 

end 

queue "/*" 

queue "##" 

if test then do 

   queue 

   call bpxwdyn "alloc dsorg(PS) recfm(F) space(15) tracks rtddn(ddn)" 

   "execio * diskw" ddn "(finis" 

   zedlmsg = "Submit gegebenenfalls selbst absetzen" 

   address ispexec "setmsg msg(isrz000)" 

   call #ediv ddn /* User Defined Function – im ISPF-Teil beschrieben */ 

   "free dd("ddn")" 

End 

else do 

   "submit * end(##)" 

End 

"delstack" 

Exit 

4.17.8.10. AKUV#TL2 – Telefonliste Variante 2 

Diese Variante zeigt, wie einzelne Datensätze über File Tailoring verarbeitet werden können. 

/* REXX * AKUV#TL2 * Erstellen Telefonliste 

*/ 

name. = left(" ",35) 

address ispexec 

"control errors return" 

"vget (ShrVars) shared" 

"vget (ztempn ztempf &ShrVars) shared" 

"tbquery &kdtab sortflds(AltSort)" 

AltSort = strip(translate(AltSort,"","()")) 

"tbsort &kdtab fields(#NNAMEU,C,A,#VNAMEU,C,A)" 

"ftopen temp" 

call Build_JCL 

call Build_Data 

SkelLine = "/*" 

"ftincl AKUV#NL" 

"FtClose" 

"tbsort &kdtab fields(&AltSort)" 

 

if Admin then do 

   zedlmsg = "Bitte gegebenenfalls selbst SUBMIT absetzen" 

   address ispexec "setmsg msg(isrz000)" 

   call #ediv "/V" /* User Defined Function – im ISPF-Teil beschrieben */ 

end 

else address tso "submit '"zTempF"'" 

exit 

Include_Data: procedure expose $ Tel. Name. 

   do §=1 to $ 

      SkelLine = Name.§ Tel.§ 

      "ftincl AKUV#NL" 

   End 

Return 

 

Build_JCL: 

   SkelLine = "//TELVERZ EXEC PGM=IEBGENER";"ftincl AKUV#NL" 

   SkelLine = "//SYSIN    DD DUMMY        ";"ftincl AKUV#NL" 

   SkelLine = "//SYSPRINT DD DUMMY        ";"ftincl AKUV#NL" 

   SkelLine = "//SYSUT2   DD SYSOUT=*     ";"ftincl AKUV#NL" 

   SkelLine = "//SYSUT1   DD *            ";"ftincl AKUV#NL" 

Return 

 

Build_Data: 

   do forever 

      "tbskip &kdtab" 

      if rc > 0 then leave 

      if space(#TELFN #TELMOB #FAX,0) = "" then iterate 

      $=0 

      if strip(#telfn)  <> "" then do 

         $=$+1;Tel.$ = "Festnetz:" #telfn 

      End 

      if strip(#telmob) <> "" then do 

         $=$+1;Tel.$ = "Mobil:   " #telmob 

      End 

      if strip(#fax) <> "" then do 

         $=$+1;Tel.$ = "Fax:     " #fax 

      End 

      name.1 = left(strip(#nname)"," strip(#vname),35,".") 

      call Include_Data 

   end 

return 

4.17.8.11. AKUV#OA – Lists der offenen Aufträge 

/* REXX * AKUV#OA * Erstellen Liste Offene Aufträge 

*/ 

parse source . rxTYP rxNAME . 

test = 1 

§ = 0 

name. = left(" ",35) 

address ispexec 

"control errors return" 

"vget (ztempn shrvars) shared" 

"vget (&shrvars) shared" 

"tbquery &auftab sortflds(AltSort)" 

AltSort = strip(translate(AltSort,"","()")) 

"tbsort &auftab fields($aufkdnr,N,A,$aufdat,N,A)" 

"tbtop &auftab" 

"tbvclear &auftab" 

$aufVKZ = "ERL" 

"tbscan &auftab arglist($aufVKZ) condlist(NE)" 

do while rc = 0 

   "tbtop &kdtab" 

   #kdnr = $aufkdnr 

   "tbscan &kdtab arglist(#kdnr) condlist(EQ)" 

   $aufdat = translate("Tt.Mm.CcJj",$aufdat,"CcJjMmTt") 

   § = § + 1 

   oa.§ = right($aufkdnr,4) left(strip(#nname)"," strip(#vname),24) , 

          $aufdatd right($artnum, 6) right($artanz,6) , 

          right($artepr,8) left($artbez,12) 

   $aufVKZ = "ERL" 

   "tbscan &auftab arglist($aufVKZ) condlist(NE)" 

End 

"ftopen temp" 

skelinfo = "Job-Anweisung für Liste Offene Aufträge" 

"ftincl AKUV#JOB" 

skelinfo = "IEBGENER für Liste Offene Aufträge" 

"ftincl AKUV#GE1" 

do $=1 to § 

   skelline = oa.$ 

   "ftincl AKUV#NL" 

End 

skelline = "/*" 

"ftincl AKUV#NL" 

"ftclose" 

"tbsort &auftab fields(&AltSort)" 

if test then do 

   zedlmsg = "JOB is 4 U – SUBMIT is yours 

   "setmsg msg(isrz000)" 

   "lminit dataid(DID) ddname(&ZTEMPN)" 

   "view dataid(&DID)" 

   "lmfree dataid(&DID)" 

End 

else do 

   address tso "submit '"ztempf"'" 

end 

exit 

4.17.8.12. AKUV#GL1 – Geburtstagsliste Variante 1 

Die Sortierung der Liste erfolgt über ein Edit Macro 

/* REXX * AKUV#GEB * Erstellen Geburtstagsliste (Sort über Edit Makro) 

*/ 

call msg "OFF" 

§ = 0 

 

address ispexec 

"control errors return" 

"vget (ShrVars)  shared" 

"vget (&ShrVars) shared" 

 

"tbtop &kdtab" 

"tbskip &kdtab" 

do while rc = 0 

   if strip(#gebdat) <> "" then do 

      §=§+1;geb.§ = #gebdat".."strip(#nname)", "strip(#vname) 

   End 

   "tbskip &kdtab" 

End 

 

address tso 

dsntemp = "'"userid()".geblist.akuvigg.temp'" 

"del" dsntemp 

"alloc dd(geblist) da("dsntemp") new catalog" , 

   "recfm(f b) lrecl(70) blksize(0) space(1 1) tracks dsorg(ps)" 

"execio * diskw geblist (stem geb. finis" 

"free dd(geblist)" 

 

SortParm = "4 5 A 1 2 A 7 10 D 13 40 A" 

address ispexec "vput SortParm shared" 

address ispexec "edit dataset(&dsntemp) macro(AKUV#EM1)" 

 

"alloc dd(sysin) dummy" 

"alloc dd(sysprint) dummy" 

"alloc dd(sysut1) da("dsntemp") shr reuse" 

if admin then generparm = "DA(*)" 

else          generparm = "SYSOUT(H)" 

"alloc dd(sysut2)" generparm 

"iebgener" 

"free dd(sysin sysprint sysut1 sysut2)" 

exit 

4.17.8.13. AKUV#EM1 – Edit Macro zum Sortieren 

/* REXX * AKUV#EM1 * Edit Macro Sort Gebliste 

*/ 

address ispexec "vget sortparm shared" 

address isredit 

"macro" 

"sort &sortparm" 

"save" 

"cancel" 

exit 

4.17.8.14. AKUV#GL2 – Geburtstagsliste Variante 2 

Die Sortierung der Liste erfolgt über eine temporäre ISPF-Tabelle 

/* REXX * AKUV#GEB * Erstellen Geburtstagsliste (Sort über TTab-Table) 

*/ 

call msg "OFF"                                                                 

 

address ispexec 

"control errors return" 

"vget (KdTab)  shared" 

 

"tbend    TTab" 

"tbcreate TTab names(Jahr Monat tag #nname #vname #gebdat) nowrite" 

"tbsort   TTab fields(Monat,N,A,Tag,N,A,Jahr,N,D,#NName,C,A,#VName,C,A)" 

"tbtop  &kdtab" 

"tbskip &kdtab" 

do while rc = 0 

   if strip(#gebdat) <> "" then do 

      parse var #gebdat Tag "." Monat "." Jahr 

      "tbadd TTab order" 

   End 

   "tbskip &kdtab" 

End 

"tbtop TTab" 

do §=1 

   "tbskip TTab" 

   if rc > 0 then leave 

   geb.§ = #gebdat".."strip(#nname)", "strip(#vname) 

end 

"tbend TTab" 

 

address tso 

"alloc dd(sysin) dummy" 

"alloc dd(sysprint) dummy" 

"alloc dd(sysut2) da(*)" 

call bpxwdyn "alloc dd(SYSUT1) lrecl(70) Space(15) tracks dsorg(PS)" 

"execio * diskw sysut1 (stem geb. finis" 

"iebgener" 

"free dd(sysin sysprint sysut1 sysut2)" 

Exit 

4.17.8.15. AKUV-KAL – Kalenderanzeige 

Ist auch nichts, das unverzichtbar wäre. Es zeigt aber, wie einfach der Umgang mit Datumsfeldern 
in REXX ist. Außerdem soll der Umgang mit einer Dynamic Area im Panel veranschaulicht werden. 

Das Kalenderprogramm kann in der Kundenverwaltung durch den Befehl KAL aufgerufen werden. 

/* REXX * AKUV#KAL * Kalenderermittlung 

*/ 

arg jahr 

if jahr = "" then jahr = word(date(),3) 

 

call set_vars 

 

oso       = ostern(jahr) 

feiertage = ftage(jahr) 

speztage  = stage(jahr) 

geburtage = gtage(jahr) 

 

call set_kalender 

 

address ispexec 

"vget zScreenD shared" 

if zScreenD >= 27 then do 

   kaldat_dup = kaldat 

   zwinttl = "< Kalender für das Jahr" jahr" >" 

   "addpop row(2) column(4)" 

   do forever 

      "display panel(AKUV#KAL) cursor(kaldat) csrpos("CSRkal")" 

      if rc > 0 then leave 

      if pos(substr(kaldat,curpos,1),"MDFS") = 0 then iterate 

      call check_tag 

      kaldat = kaldat_dup 

   end 

   "rempop" 

End 

else do 

   zedlmsg = "Für die Kalenderanzeige muss die Screensize" , 

             "mindestens 27 Zeilen hoch sein" 

   "setmsg msg(isrz000)" 

End 

 

Exit 

 

check_tag: 

   kaldat = overlay("°",kaldat,curpos) 

   kaldat = space(translate(kaldat," ","0123456789^{}\]§²"),0) 

   ?tag   = pos("°",kaldat) 

   wotag  = word(T_D,wordpos(date("W",datum.?tag,"B"),T_E)) 

   parse value date("S",datum.?tag,"B") with 1 jahr 5 mon 7 tag 

   select 

      when ftag.?tag <> ""     then msg = "("ftag.?tag")" 

      when left(wotag,1) = "S" then msg = "(Wochenende)" 

      otherwise                     msg = "("?tag")" 

   end 

   rest = length(substr(kaldat,?tag+1)) 

   zedlmsg = wotag"," tag"."mon"."jahr msg 

   zedlmsg = center(" "zedlmsg "- noch" rest "Tage. ",65,"*") 

   "setmsg msg(isrz000)" 

   CSRkal = curpos 

Return 

 

ostern: procedure 

   arg jahr 

   /******************************************************************* 

   * Bewegliche Feiertage sind abhängig von Ostern. Dies ist abhängig * 

   * vom 1. Vollmond im Frühling. Berechnung nach Gauß'scher Formel   * 

   *******************************************************************/ 

   a=jahr//19 

   b=jahr//4 

   c=jahr//7 

   d=((19*a+24)//30) 

   e=((2*b+4*c+6*d+5)//7) 

   f=22+d+e 

   if f = 56 & d=28 & e=6 & a > 10 then f=49 

   if f <= 31 then m="03" 

   else do 

      f=f-31 

      m="04" 

   end 

return date("B",jahr !! m !! right(f,2,0),"S") 

 

ftage: 

   arg jahr 

   /******************************************************************* 

   * Feiertage                                                        * 

   *******************************************************************/ 

   feiertage = "" !!           ,   /* Wortkette Feiertage BASE-Fmt */ 

      oso-2                    ,   /* Karfreitag                   */ 

      oso                      ,   /* Ostersonntag                 */ 

      oso+1                    ,   /* Ostermontag                  */ 

      oso+39                   ,   /* Jesus Skydrive               */ 

      oso+49                   ,   /* Pfingstsonntag               */ 

      oso+50                   ,   /* Pfingstmontag                */ 

      oso+60                   ,   /* Frohnleichnam                */ 

      date("B",jahr"0101","S") ,   /* Neujahr                      */ 

      date("B",jahr"0106","S") ,   /* Heilig 3 König               */ 

      date("B",jahr"0501","S") ,   /* Tag der Arbeit               */ 

      date("B",jahr"1003","S") ,   /* Tag der Einheit              */ 

      date("B",jahr"1101","S") ,   /* Allerheiligen                */ 

      date("B",jahr"1225","S") ,   /* 1.Weihnachtstag              */ 

      date("B",jahr"1226","S") ,   /* 2.Weihnachtstag              */ 

      "" 

   interpret "FT."word(feiertage,1) " = 'Karfreitag'" 

   interpret "FT."word(feiertage,2) " = 'Ostersonntag'" 

   interpret "FT."word(feiertage,3) " = 'Ostermontag'" 

   interpret "FT."word(feiertage,4) " = 'Christi Himmelfahrt'" 

   interpret "FT."word(feiertage,5) " = 'Pfingstsonntag'" 

   interpret "FT."word(feiertage,6) " = 'Pfingstmontag'" 

   interpret "FT."word(feiertage,7) " = 'Frohnleichnam'" 

   interpret "FT."word(feiertage,8) " = 'Neujahr'" 

   interpret "FT."word(feiertage,9) " = 'Heilig drei König'" 

   interpret "FT."word(feiertage,10)" = 'Tag der Arbeit'" 

   interpret "FT."word(feiertage,11)" = 'Tag der Enheit'" 

   interpret "FT."word(feiertage,12)" = 'Allerheiligen'" 

   interpret "FT."word(feiertage,13)" = '1. Weihnachtstag'" 

   interpret "FT."word(feiertage,14)" = '2. Weihnachtstag'" 

return feiertage 

 

stage: 

   arg jahr 

   /***************************************************************** 

   * Spezielle Tage (Fasching und nicht bundesweite Feiertage       * 

   *****************************************************************/ 

   speztage = "" !!            ,   /* Wortkette Feiertage BASE-Fmt */ 

      oso-49                   ,   /* Faschingssonntag             */ 

      oso-48                   ,   /* Rosenmontag                  */ 

      oso-47                   ,   /* Faschingsdienstag            */ 

      oso-46                   ,   /* Aschermittwoch               */ 

      date("B",jahr"0815","S") ,   /* Maria Himmelfahrt            */ 

      date("B",jahr"1224","S") ,   /* Hl. Abend                    */ 

      date("B",jahr"1231","S") ,   /* Sylvester                    */ 

      "" 

   interpret "FT."word(speztage,1) " = 'Faschingssonntag'" 

   interpret "FT."word(speztage,2) " = 'Rosenmontag'" 

   interpret "FT."word(speztage,3) " = 'Faschingsdienstag'" 

   interpret "FT."word(speztage,4) " = 'Aschermittwoch'" 

   interpret "FT."word(speztage,5) " = 'Maria Himmelfahrt'" 

   interpret "FT."word(speztage,6) " = 'Heilig Abend'" 

   interpret "FT."word(speztage,7) " = 'Sylvester'" 

return speztage 

 

gtage: 

   arg jahr 

   /***************************************************************** 

   * Geburtstage festlegen                                          * 

   *****************************************************************/ 

   gebtage = "" !!             ,   /* Wortkette Geburtage BASE-Fmt */ 

      date("B",jahr"0917","S") ,   /* Karl-Heinz Wittemann         */ 

      date("B",jahr"0407","S") ,   /* MVS ward geboren             */ 

      "" 

   interpret "FT."word(gebtage,1) "='Heinzilein wird" jahr-1954"'" 

   interpret "FT."word(gebtage,2) "='Der Mainframe wird" jahr-1964"'" 

return gebtage 

 

set_vars: 

   ft. = 

   tab. = 

   Mon_D  = "Jan Feb Mär Apr Mai Jun Jul Aug Sep Okt Nov Dez" 

   T_E = "Monday Tuesday  Wednesday Thursday   Friday  Saturday Sunday" 

   T_D = "Montag Dienstag Mittwoch  Donnerstag Freitag Samstag  Sonntag" 

   basedat = date("B",jahr"0101","S") 

return 

 

set_kalender: 

   do _cnt = 1 

      parse value date("S",basedat,"B") with 1 jjjj 5 mm 7 tt 

      if jahr < jjjj then leave 

      dow  = date("W",basedat,"B") 

      dowd = left(word(T_D,wordpos(DOW,T_E)),1) 

      select 

         when wordpos(basedat,speztage) > 0   then tmp = "§" 

         when wordpos(basedat,geburtage) > 0  then tmp = "²" 

         when wordpos(basedat,feiertage) > 0  then tmp = "{" 

         when dowd = "S" 

         otherwise do 

            if basedat = date("B") then tmp = "\" 

            else tmp = "]" 

         end 

      end 

      datum._cnt = basedat 

      ftag._cnt  = ft.basedat 

      tab.mm = tab.mm !! tmp !! dowd 

      basedat = basedat + 1 

   end 

   kaldat = 

   do i=1 to 12 

      i=right(i,2,0) 

      kaldat = kaldat !! "^"i !! left(tab.i"^ ",65) copies(" ",65) 

   end 

   csrkal = pos("\",kaldat)+1 



Inside z/OS Seite 318 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

return 

4.17.8.16. AKUV#FUN – ASCII-Signaturen,  Aufruf durch CMD-Table 

Das folgende Programm ist in der Schriftgröße um 1 Punkt kleiner, damit die Darstellung  (Zeilenum-
brüche) vernünftig ist. 

/* REXX 

*/ 

arg what . 

ScreenD = strip(sysvar("SYSLTERM")) 

signatures = "Smily Coffee Welcome Killroy" 

upper what signatures 

do §=1 to words(Signatures) 

   Sign = word(signatures,§) 

   if left(Sign,length(what)) = what & what <> "" then do 

      what = sign 

      leave 

   end 

end 

if wordpos(what,Signatures) = 0 then do 

   zedlmsg = signatures 

   address ispexec "setmsg msg(isrz000)" 

   exit 

end 

interpret "call" what 

parse var result linesize data 

"cls" 

do (screend - length(data) % linesize) % 2 

   SAY 

End 

do while data <> "" 

   parse var data 1 dataline =(linesize) data 

   say center(dataline,linesize()) 

end 

say;say;say 

exit 

 

Smily: return 75 , 

 "                         oooo$$$$$$$$$$$$oooo                             "!!, 

 "                      oo$$$$$$$$$$$$$$$$$$$$$$$$o                         "!!, 

 "                   oo$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$o         o$   $$ o$   "!!, 

 "   o $ oo        o$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$o       $$ $$ $$o$  "!!, 

 "   $ $ '$      o$$$$$$$$$    $$$$$$$$$$$$$    $$$$$$$$$o       $$$o$$o$   "!!, 

 "'$$$$$$o$     o$$$$$$$$$      $$$$$$$$$$$      $$$$$$$$$$o    $$$$$$$$    "!!, 

 "  $$$$$$$    $$$$$$$$$$$      $$$$$$$$$$$      $$$$$$$$$$$$$$$$$$$$$$$    "!!, 

 "  $$$$$$$$$$$$$$$$$$$$$$$    $$$$$$$$$$$$$    $$$$$$$$$$$$$$  '''$$$      "!!, 

 "   '$$$''''$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$     '$$$     "!!, 

 "    $$$   o$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$     '$$$o   "!!, 

 "   o$$'   $$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$       $$$o  "!!, 

 "   $$$    $$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$' '$$$$$$ooooo$$$$o"!!, 

 "   $$$oooo$$$$$  $$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$   o$$$$$$$$$$$$$$$$"!!, 

 "  $$$$$$$$'$$$$   $$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$     $$$$             "!!, 

 "            $$$$    '$$$$$$$$$$$$$$$$$$$$$$$$$$$$'      o$$$              "!!, 

 "            '$$$o     '''$$$$$$$$$$$$$$$$$$'$$'         $$$               "!!, 

 "              $$$o          '$$''$$$$$$''''           o$$$                "!!, 

 "               $$$$o                 oo             o$$$'                 "!!, 

 "                '$$$$o      o$$$$$$o'$$$$o        o$$$$                   "!!, 

 "                  '$$$$$oo     ''$$$$o$$$$$o   o$$$$''                    "!!, 

 "                     ''$$$$$oooo  '$$$o$$$$$$$$$'''                       "!!, 

 "                        ''$$$$$$$oo $$$$$$$$$$                            "!!, 

 "                                ''''$$$$$$$$$$$                           "!!, 

 "                                    $$$$$$$$$$$$                          "!!, 

 "                                     $$$$$$$$$$'                          "!!, 

 "                                       $$$$$$°                            "    

 

Coffee: return 34 , 

 "        /.                       " !! , 

 "       / !                       " !! , 

 "      !  !                       " !! , 

 "      /  /                       " !! , 

 "     !  !                        " !! , 

 "    __!__!__________________     " !! , 

 "    |                      |__   " !! , 

 "    '  Wolle Kaffe trinke  '  !  " !! , 

 "     \                    /   !  " !! , 

 "      \   wege miede ?   /___/   " !! , 

 "       \                /        " !! , 

 "________\______________/________ " !! , 

 "        ________________         " 

 

Welcome: return 54 , 

 "                                                     " !! , 

 "      __                 ___                 __      " !! , 

 "  ___(  \             .--   --.             /  )___  " !! , 

 " (___    \____       /  (O O)  \       ____/    ___) " !! , 

 "(___          \-----(     )     )-----/          ___)" !! , 

 " (___    ____________\  \___/  /____________    ___) " !! , 

 "   (____/              -.___.-              \____)   " !! , 

 "                                                     " 

 

Killroy: return 54 , 

 "                                                     " !! , 

 "                         ___                         " !! , 

 "                        .   .                        " !! , 

 "                       ( o o )                       " !! , 

 "+-----------------.oooO--(_)--Oooo.-----------------+" !! , 

 "|                                                   |" !! , 

 "|      Heinzi ist   .oooO  der Allerbeste           |" !! , 

 "|                   (   )   Oooo.                   |" !! , 

 "+--------------------\ (----(   )-------------------+" !! , 

 "                      \_)    ) /                     " !! , 

 "                            (_/                      " !! , 

 "                                                     " 

4.17.9. Die Panels 

4.17.9.1. AKUVIGG – Das hierarchisch oberste Auswahlpanel 

Wenn alle Vorbereitungen getroffen sind, alle Bibliotheken zugewiesen, alle Tabellen geöffnet wur-
den, kann die Auswahlmaske angezeigt werden. Hier das entsprechende Selection Panel. 

)ATTR default(%+_) 

   + TYPE(TEXT)   COLOR(TURQ) 

   % TYPE(TEXT)   COLOR(YELLOW) 

   # TYPE(TEXT)   COLOR(RED) 

   $ TYPE(TEXT)   COLOR(WHITE)  HILITE(USCORE) 

   µ TYPE(TEXT)   COLOR(WHITE) 

   ² TYPE(TEXT)   COLOR(PINK)   HILITE(USCORE) 

   ] TYPE(TEXT)   COLOR(BLUE) 

   [ TYPE(TEXT)   COLOR(GREEN)  HILITE(REVERSE) 

   â AREA(SCRL)    EXTEND(ON) 

   ³ area(dynamic) extend(off) 

   Á type(DATAOUT) color(white) 

   é type(char)    color(green) 

   ó type(char)    color(red) 

   ú type(char)    color(blue) 

   í type(char)    color(turq) 

   ö type(char)    color(pink) 

   ü type(char)    color(yellow) 

)BODY expand(\\) 

+\ \µA K U V I G G+\ \ 

$\ \µPrimary Option Menue$\ \ 

+Auswahl ==>_ZCMD 

+ 

                 ³MainPdat,MainPshd                              ³ 

                 ³                                               ³ 

                 ³                                               ³ 

                 ³                                               ³ 

                 ³                                               ³ 

                 ³                                               ³ 

+ 

+\ \µAuftrags- und KUndenVerwaltung+\ \ 

+\ \µInter Galakitsch Gut+\ \ 

âPanBody 

)AREA PanBody 

%µDaten bearbeiten+ 

%        K   (1)  [Kunde      +Bearbeitung Kundentabelle 

%        A   (2)  [Auftrag    +Bearbeitung Auftragstabelle 

% 

%µRechnungstellung+ 

%        RE  (11) [Faktura    +Einzel-/Gesamtrechnungslauf 

% 

%µListausgaben    + 

%        AL  (21) [KD-List    +Kundenliste alphabetisch 

%        TL  (22) [TelVerz    +Telefonverzeichnis 

%        OA  (23) [Aufträge   +Liste offener Aufträge 

% 

%        GEB (24) [GebTag     +Geburtstagsliste 

% 

%       ²                                                           + 

% 

%        X   (99) [Ende       +&pendtext 

% 

% 

#\ \&ZUser - Datum: &tt..&mm..&ZSTDYEAR - Zeit: &ZTIME \ \ 

)INIT 

   &ZPRIM = YES        /* ALWAYS A PRIMARY OPTION MENU      */ 

   .cursor = ZCMD 

   &jj   = trunc(&ZDate,'/') 

   &rest = .trail 

   &mm   = trunc(&rest,'/') 

   &tt   = .trail 

   vget (MainPdat MainPshd PEndText) shared 

)PROC 

   &ZSEL = TRANS(&ZCMD 

                 K,'cmd(%AKUV#KUN)' 

                 1,'cmd(%AKUV#KUN)' 

                 A,'cmd(%AKUV#AUF)' 

                 2,'cmd(%AKUV#AUF)' 

                RE,'cmd(%AKUV#SRE)' 

                11,'cmd(%AKUV#SRE)' 

                AL,'cmd(%AKUV#ALI)' 

                21,'cmd(%AKUV#ALI)' 

                TL,'cmd(%AKUV#TL1)' 

                22,'cmd(%AKUV#TL2)' 

                OA,'cmd(%AKUV#OA)' 

                23,'cmd(%AKUV#DMY "Liste <offene Aufträge>")' 

               GEB,'cmd(%AKUV#GL1)' 

                24,'cmd(%AKUV#GL2)' 

                 X,'EXIT' 

                99,'EXIT' 

               ' ','cmd(%AKUV#COL)' 

                 *,? 

                 ) 

)END 

 

Hier kann nun die Auswahl innerhalb des Dialoges erfolgen. 

4.17.9.2. AKUV#KDT – Das Tabellenpanel der Kundenverwaltung 

Das Tabellenpanel erlaubt die Filtrierung der Anzeige, indem in die Felder oberhalb der Spalten-
überschriften Werte Eingetragen werden. Enthalten alle Eingabefelder einen Stern(*), werden alle 
Datensätze angezeigt. Wird beispielsweise im Namensfeld Au* und im PLZ-Feld 8*eingetragen, wer-
den nur Sätze gezeigt, bei denen der Name mit Au und die Postleitzahl mit 8 beginnen. 

)ATTR default(%+_) 

   ~ TYPE(INPUT)   caps(ON)       padc('.') 

   @ TYPE(OUTPUT)  caps(OFF)      color(&farbe1) just(right) 

   \ TYPE(OUTPUT)  caps(OFF)      color(&farbe2) 

   µ TYPE(OUTPUT)  caps(OFF)      color(&farbe3) 

   ° TYPE(OUTPUT)  caps(OFF)      color(&farbe4) 

   á TYPE(OUTPUT)  caps(OFF)      color(&farbe5) 

   ` TYPE(OUTPUT)  caps(OFF)      color(&farbe6) 

   ² TYPE(TEXT)    hilite(uscore) color(white) 

   ³ TYPE(INPUT)   caps(OFF)      color(red) 

   ] AREA(DYNAMIC) 

   } TYPE(DATAOUT) color(white) 

   { TYPE(DATAOUT) color(blue) 

)BODY expand(//) 

+----------<%Bearbeitung Kundentabelle+>/-/ 

+COMMAND =>_ZCMD / /+SCROLL =>_SAMT+ 

+ 

+PrimCMDs:]PCMDS                                                      ] 

+LineCMDs:]LCMDS                                                      ] 

%|+ 

%|+     ³Snname         ³Svname      ³Splz ³Sort               ³Sstr 

%V²KDNR ²Name           ²Vorname     ²PLZ  ²Ort                ²Straße 

)MODEL clear(LCMD) rows(&KDSHOW) 

~Z@#KDNR\#NNAME         µ#VNAME      °#PLZ á#ORT               `#STR 

§ 

)INIT 

   .zvars = '(LCMD)' 

   if (&SAMT = &Z)      &SAMT = CSR 

   if (&KDSHOW ^= SCAN) &KDSHOW = ALL 

)PROC 

   if (ver (&lcmd,listv,&kdlcmds)) 

   else 

      &zedlmsg = 'Ungültiges Zeilenkommando' 

      .MSG = ISRZ000 

)END 

4.17.9.3. AKUV#KDD – Das Datenpanel zu den Kundensätzen 

Ich bitte die verschachtelte Abfragelogik im )PROC-Teil zu entschuldigen. Die Absicht war, alle Va-
riationen einmal zu zeigen – auch die weniger guten. 

)ATTR default($+_) 

   ] type(text)   color(BLUE) 

   ~ type(output) color(YELLOW) caps(OFF)    padc('.') 

   ° type(output) color(BLUE)   caps(OFF)    padc('.') 

   ² AREA(DYNAMIC) 

   } TYPE(DATAOUT) color(WHITE) 

   { TYPE(DATAOUT) color(BLUE) 

)BODY expand(//) window(50,19) 

+Command ==>_ZCMD 

+ 

]Kunden-Nr.>~#KDNR+ 

]Kunden-Gr.>°Z+                        Mussfeld 

] 

]Name......>°#NNAME               +    Mussfeld 

]Vorname...>°#VNAME               +    Mussfeld 

]PLZ.......>°#PLZ +                    Mussfeld 

]Ort.......>°#ORT                     +Mussfeld 

]Strasse...>°#STR                     +Mussfeld 

] 

]Telefon...>°#TELFN           +        Kannfeld 

]Mobil.....>°#TELMOB          +        Kannfeld 

]Fax.......>°#FAX             +        Kannfeld 

] 

]Geb.-Datum>°#GEBDAT   +(tt.mm.jjjj)   Kannfeld 

] 

²PANMSG                                       ² 

)INIT 

   .zvars = '(#KDGR)' 

   &ZCmd = &Z 

   if (&Lcmd = I,U) 

      .AttrChar(°) = 'type(INPUT) color(GREEN)' 

      .AttrChar(]) = 'type(TEXT)  color(TURQ)' 

)REINIT 

   refresh (*) 

   .attr(.cursor) = 'color(red)' 

)PROC 

   if (ver (&#KDGR,NB,RANGE,0,4)) 

      if (ver (&#NNAME,NB)) 

         if (ver (&#VNAME,NB)) 

            if (ver (&#PLZ,NB,NUM)) 

               if (ver (&#ORT,NB)) 

                  if (ver (&#STR,NB)) 

                     if (ver (&#GEBDAT,PICT,'99.99.9999')) 

                     else 

                        &zedlmsg = 'Datumsformat muss tt.mm.jjjj sein' 

                        .MSG = ISRZ000 

                  Else 

                     &zedlmsg = 'Straße ist ein Pflichtfeld' 

                     .MSG = ISRZ000 

               Else 

                  &zedlmsg = 'Ort ist ein Pflichtfeld' 

                  .MSG = ISRZ000 

            Else 

               &zedlmsg = 'PLZ ist ein numerisches Pflichtfeld' 

               .MSG = ISRZ000 

         Else 

            &zedlmsg = 'Vorname ist ein Pflichtfeld' 

            .MSG = ISRZ000 

      Else 

         &zedlmsg = 'Nachname ist ein Pflichtfeld' 

         .MSG = ISRZ000 

   Else 

      &zedlmsg = 'Nur Werte zwischen 0 und 4' 

      .MSG = ISRZ000 

   &ZWINTTL = &Z 

)HELP 

   field(#KDGR) panel(AHLP#KGR) 

)END 

 

4.17.9.4. AHLP#KGR – Fieldhelp für die Kundengruppe 

)BODY default($+_) window(16,7) 

+ 

+0 = Netto 

+1 =  5% Rabatt 

+2 = 10% Rabatt 

+3 = 20% Rabatt 

+4 = 25% Rabatt 

)INIT 

&ZWINTTL = '<Help KuGru>' 

)PROC 

&ZWINTTL = &z 

)END 

4.17.9.5. AKUV#AUT – Das Tabellenpanel zur Auftragstabelle 

Die Anzeige der Tabelle wartet mit einem Dynamic Modelset auf. Das heißt, der Aufbau der Tabel-
lenzeilen wird über eine Variable geregelt, die je nach Bedarf im Programm aufgebaut werden kann. 
Die entsprechend dazu passende Spaltenüberschrift wird in einer Dynamic Area angezeigt. 

)ATTR default(%+_) 

   { TYPE(INPUT)   caps(OFF)      color(RED) 

   ~ TYPE(INPUT)   caps(ON)       padc('.') 

   } TYPE(TEXT)    color(YELLOW)  padc('.') 

   \ TYPE(OUTPUT)  caps(OFF)      color(TURQ) 

   µ TYPE(OUTPUT)  caps(OFF)      color(TURQ)   just(right) 

   [ area(dynamic) 

   ² TYPE(DATAOUT) hilite(uscore) color(white) 

   ³ TYPE(DATAOUT) color(white) 

   ^ TYPE(DATAOUT) color(blue) 

)BODY expand(//) 

+----------<%Bearbeitung Auftragstabelle+>/-/ 

+COMMAND =>{ZCMD / /+SCROLL =>_SAMT+ 

+ 

+PrimCMDs:%I+(Insert) %SCAN+parm  %R+(ReorgTable) %1|2|3+(Ansichten) 

+LineCMDs:[lineCM                                                   [ 

%|+ 

%|}Ansicht sortiert nach &sortiert 

%|+ 

[HEADER 

[  

)MODEL clear(LCMD) rows(&AUFSHOW) 

&MODSET 

)INIT 

   .zvars = '(&ZFIELD)' 

   if (&SAMT = &Z)       &SAMT = CSR 

   if (&aufSHOW ^= SCAN) &aufSHOW = ALL 

)PROC 

   ver (&lcmd,listv,&aufLCMDS) 

)END 

4.17.9.6. AKUV#AUD – Das Datenpanel zu den Auftragssätzen 

Dies ist das Standardpanel für den "gemeinen Benutzer". 

)ATTR default(%+_) 

   ] type(text)   color(TURQ) 

   ° type(INPUT)  color(YELLOW) caps(OFF)    padc('.') 

   ² AREA(DYNAMIC) 

   } TYPE(DATAOUT) color(WHITE) 

   { TYPE(DATAOUT) color(BLUE) 

)BODY expand(//) window(50,13) 

+Command ==>_ZCMD 

+ 

] 

²PANMSG1                                      ² 

] 

]Auftrags-KdNr>°Z     + 

]Anzahl.......>°Z     + 

]Artikelnummer>°Z     + 

]Artikel-Bez..>°$ARTBEZ                       + 

]Einzelpreis..>°$ARTEPR+ 

] 

²PANMSG2                                      ² 

)INIT 

   &ZCMD = &Z 

   .zvars = '($AUFKDNR $ARTANZ $ARTNUM)' 

)REINIT 

   .attr(.cursor) = 'color(red)' 

)PROC 

   ver (&$ARTANZ,NB,NUM) 

   ver (&$ARTNUM,NB,NUM) 

   ver (&$ARTBEZ,NB) 

   ver (&$ARTEPR,NB,ENUM) 

   &ZWINTTL = &Z 

)END 

4.17.9.7. AKUV#AUG – Auftrags-Datenpanel für den Administrator 

Hier sind mehr als die "normalen" Felder änderbar. Wer Admin-Rechte hat, kann ALLES ändern. Es 
stellt sich die Frage, ob man das in EINEM Panel hätte realisieren können. Die Antwort ist: SELBST-
VERSTÄNDLICH. Aber als "Anschauungsmaterial" sollte es auch vernünftig lesbar bleiben. 

Eines der größeren Probleme ist, dass die Größe der Window-Definition über Variablen nicht erlaubt 
ist. Der geschätzte Realisierungsaufwand seitens der IBM läge meines Erachtens im Minutenbe-
reich. Auch das kann man in den Griff bekommen, allerdings würde das hier zu weit führen. 

)ATTR default(%+_) 

   ] type(text)   color(TURQ) 

   ° type(INPUT)  color(YELLOW) caps(OFF)    padc('.') 

   ² AREA(DYNAMIC) 

   } TYPE(DATAOUT) color(WHITE) 

   { TYPE(DATAOUT) color(BLUE) 

)BODY expand(//) window(50,15) 

+Command ==>_ZCMD 

+ 

] 

²PANMSG1                                      ² 

] 

]Auftrags-KdNr>°Z     + 

]Anzahl.......>°Z     + 

]Artikelnummer>°Z     + 

]Artikel-Bez..>°$ARTBEZ                       + 

]Einzelpreis..>°$ARTEPR+ 

]Verarb.-KZ...>°Z  + 

]Auftragsdatum>°$aufdat + 

] 

²PANMSG2                                      ² 

)INIT 

   &ZCMD = &Z 

   if (&ZWINTTL = &Z) &ZWINTTL = 'Paneltest' 

   .zvars = '($AUFKDNR $ARTANZ $ARTNUM $aufVKZ)' 

)REINIT 

   .attr(.cursor) = 'color(red)' 

)PROC 

   ver (&$ARTANZ,NB,NUM) 

   ver (&$ARTNUM,NB,NUM) 

   ver (&$ARTBEZ,NB) 

   ver (&$ARTEPR,NB,ENUM) 

   &ZWINTTL = &Z 

)END 

4.17.9.8. AKUV#SRE – Auswahl für Rechnungen 

Ruft das Programm für die Rechnungstellung auf. Abhängig von der Bedienerauswahl werden un-
terschiedliche Parameter übergeben, anhand derer entschieden wird, ob eine Einzelrechnung oder 
ein monatlicher Gesamt-Rechnungslauf durchgeführt werden soll. 

)ATTR default(%+_) 

# TYPE(INPUT)  COLOR(YELLOW) PADC('.') 

[ TYPE(TEXT)   INTENS(HIGH) COLOR(GREEN)  HILITE(REVERSE) 

~ TYPE(INPUT)  INTENS(NON) 

)BODY expand(\\) window(45,10) 

+ 

%Auswahl ===>_ZCMD 

+ 

+ 

%  ER +Einzelrechnung KdNr:#Z     + 

%  MR +Monatsrechnung Monat#Z     +(jjjjmm) 

% 

% 

%  END+Rückkehr ins AKUVIGG-Grundmenü 

)INIT 

   .cursor = ZCMD 

   .zvars  = '(KDNR REMON)' 

   &kdnr   = &z 

   &REMON  = '&ZSTDYEAR&ZMONTH' 

)PROC 

   if (&zcmd = ER) 

      ver (&KDNR,nb,num) 

   if (&zcmd = MR) 

      &rjahr = trunc(&remon,4) 

      &rmon  = .trail 

      ver (&Rjahr,nb,range,2007,2021) 

      ver (&Rmon,nb,range,01,12) 

   &ZSEL = TRANS(&ZCMD 

                ER,'cmd(%AKUV#RE &kdnr)' 

                MR,'cmd(%AKUV#RE &remon)' 

               ' ',' ' 

                 X,'EXIT' 

                 *,'?' ) 

)END 

4.17.10. Die Skeletons 

4.17.10.1. AKUV#JOB – Skeleton für eine Job-Anweisung 

Da die JOB-Anweisung möglichst vielseitig verwendbar sein soll, werden hier eine Reihe von Vari-
ablen abgefragt. Haben diese einen gültigen Wert, wird dieser übernommen. Auf diese Weise kann 
das Aussehen der Anweisung durch das Programm beeinflusst werden, das sie aufruft. 

)REXX JOBNAME,ACTNUM,JOBINFO,CLASS,MSGCLASS,MSGLEVEL,NOTIFY,REGSIZE,   ? 

      TIME,SCHENV 

IF STRIP(JOBNAME)  = "" THEN JOBNAME  = USERID()"#" 

IF STRIP(ACTNUM)   = "" THEN ACTNUM   = "????" 

IF STRIP(JOBINFO)  = "" THEN JOBINFO  = USERID() 

IF STRIP(CLASS)    = "" THEN CLASS    = "A" 

IF STRIP(MSGLEVEL) = "" THEN MSGLEVEL = "(1,1)" 

IF STRIP(MSGCLASS) = "" THEN MSGCLASS = "H" 

IF STRIP(REGSIZE)  = "" THEN REGSIZE  = "0K" 

IF STRIP(TIME)     = "" THEN TIME     = 1439 

IF STRIP(NOTIFY)   = "" THEN NOTIFY   = USERID() 

MSGLEVEL = STRIP(TRANSLATE(MSGLEVEL,"  ","()")) 

)ENDREXX 

//&JOBNAME JOB (&ACTNUM),'&JOBINFO', 

//         CLASS=&CLASS, 

//         MSGLEVEL=(&MSGLEVEL), 

)SEL &SCHENV NE &Z 

//         SCHENV=&SCHENV, 

)ENDSEL 

//         NOTIFY=&NOTIFY, 

//         MSGCLASS=&MSGCLASS 

//* 

//* SKEL: AKUV#JOB  - INCLUDED DURCH &RXNAME &ZTIME 

//*                   &SKELINFO 

//* 

)SEL &LPP NE &Z 

/*JOBPARM LINECT=&LPP 

)ENDSEL 

4.17.10.2. AKUV#GE1 – Skeleton für IEBGENER 

Auch das Skeleton für den IEBGENER kann sowohl mit einer Eingabedatei, als auch mit Instream-
Daten benutzt werden. Da hier im rufenden Programm die Variable DSNUT1 nicht gefüllt wurde, 
wird eine Instream-Verarbeitung durchgeführt. 

//******************************************************************* 

//* 

//* SKEL: AKUV#GE1  - INCLUDED DURCH &RXNAME &ZTIME 

//*                   &SKELINFO 

//******************************************************************* 

//IEBGENER EXEC PGM=IEBGENER 

//SYSIN    DD DUMMY 

//SYSPRINT DD SYSOUT=* 

//SYSUT2   DD SYSOUT=* 

)IF &DSNUT1 NE &Z THEN )DO 

//SYSUT1   DD DISP=SHR,DSN=&DSNUT1 

)ENDDO 

)ELSE )DO 

//SYSUT1   DD * 

)ENDDO 

4.17.10.3. AKUV#ADT – Skeleton für Verarbeitung der Adressen 

Die Daten, welche von IEBGENER Instream verarbeitet werden sollen, liefert das Skeleton über 
eine Tabellenverarbeitung. 

)DEFAULT )&?!<§> 

)TB 25A 32A 

)SET RECCNT  = 99 

)SET PAGECNT = 0 

)SET DATUM   = 00.00.0000 

)REXX DATUM 

datum = translate("Tt.Mm.CcJj",date("S"),"CcJjMmTt") 

)ENDREXX 

)DOT &KDTAB 

)SEL &RECCNT > 8 

)SET PAGECNT = &PAGECNT + 1 

)SET RECCNT  = 0 

***************** Kundenliste Gesamtübersicht AKUVIGG ***************** 

                  Stand vom: &datum   -   Seite &pagecnt 

)BLANK 2 

Name                     PLZ    ORT 

   Vorname                      Straße 

                                Telefon Festnetz 

                                Telefon mobil 

                                Telefax 

=======================  =====  ======================================== 

)ENDSEL 

)SET RECCNT = &RECCNT + 1 

)SET BLANK = 0 

)REXX blank,#telfn,#telmob,#fax 

blank = blank + (strip(#telfn)="")+(strip(#telmob)="")+(strip(#fax)="") 

)ENDREXX 

&#NNAME ! &#PLZ ! &#ORT 

   &#VNAME ! ! &#STR 

)IF &#TELFN NE &Z THEN )DO 

! ! Festnetz: &#TELFN 

)ENDDO 

)SEL &#TELMOB NE &Z 

! ! Mobil:    &#TELMOB 

)ENDSEL 

)SEL &#FAX NE &Z 

! ! Telefax:  &#FAX 

)ENDSEL 

! ! ========================== 

)BLANK &BLANK 

)ENDDOT 

)CM ******************************************************************* 

)CM Erstellen der Alphaliste aus der Kundentabelle 

)CM 

)CM 29.01.2021 Wittemann 

)CM tt.mm.jjjj Name 

)CM            Grund der Änderung  

4.17.10.4. AKUV#NL – Skeleton zur Verarbeitung einer beliebigen Zeile 

Es muss nicht immer viel sein, um brauchbar zu sein. 

&SKELLINE 

5. REXX und SDSF 

Die Kombination von REXX und SDSF erlaubt heute sehr viele Möglichkeiten, welche vor z/OS 2.3 
gar nicht, oder nur mit einem entsprechenden Aufwand und sehr unkomfortabel möglich war. Mitt-
lerweile kann im Grunde jedes SDSF-Panel via REXX aufgerufen werden und SDSF befüllt in REXX 
je Spalte einen Stem analog zur Spaltenüberschrift. Die Abarbeitung kann in einer simplen Iteration 
erfolgen. 

Einziges Hemmnis ist das Wissen über die verschiedenen Stems, welche abhängig vom aufgerufe-
nen Panel befüllt werden. Dafür wird etwas später eine kleine Routine gezeigt, welche die Namen 
der Stems zu den verschiedenen Panels ermittelt und anzeigt. 

Der Umfang der Panels nimmt laufend zu. Alleine beim Wechsel von z/OS 2.3 nach 2.4 kamen über 
15 neue hinzu. Der Umfang der Nutzung für den einzelnen ist nicht vorhersagbar, da in den meisten 
Unternehmen via RACF gesteuert wird, welche Mitarbeiter welche Panels angeboten bekommen. 
Diese Einschränkungen wollen wir hier nicht weiter berücksichtigen. 

Für die umfangreicheren Abfragemöglichkeiten über die Panels DA, ST und H stehen natürlich die 
Befehle PREFIX und OWNER zur Verfügung, wobei REXX-seitig entsprechende Variablen zu füllen 
sind. Viel besser ist aber ein entsprechendes Filtering, welches im SDSF über den FILTER-Befehl 
erfolgen kann, in REXX aber ebenfalls im Befüllen einer entsprechenden Variablen erfolgt. Gerade 
das Filtering wird von vielen Mitarbeitern sehr wenig benutzt, in der Regel, weil es nicht gekannt 
wird. 

Die Schnittstelle für SDSF im Batch wird an dieser Stelle auch nicht mehr erwähnt, da sie heute in 
der Regel nicht mehr in der Form Anwendung findet, wie es vor zOS 1.12 angesagt war. Ein "ab-
schreckendes Beispiel" ist in diesem Kapitel aber zu finden. Heute würde man im Batch über das 
IKJEFTxx ein REXX-Programm starten, dem beim Zugriff auf das SDSF letztlich egal wäre, ob es 
im Foreground oder im Background läuft. 

Dinge wie beispielsweise die Änderung von Formularkennzeichen oder Vorschubsteuerung, bzw. 
Zeichensatz im Drucker bei Ausgaben im O(utput)-Panel werden hier nicht gezeigt. Wer aber derart 
exotisches benötigt, kann das selbstverständlich auch bewerkstelligen. Beispiele hierfür sind im 
IBM-Knowledge-Center zu finden. Gegebenenfalls hilft auch das Internet weiter. 

Als Literaturempfehlung möchte ich folgende erwähnen: 

Sehr gutes Redbook zum Thema "Implementing REXX Support in SDSF" oder   
das IBM-Manual "z/OS SDSF User's Guide" 

5.1. ISFCALLS (Laden des Environments) 

Damit die Schnittstelle zu SDSF nutzbar ist, muss diese zunächst geladen werden: 

/* REXX * Demo SDSF 

*/ 

call msg "OFF" 

if IsfCalls("ON") <> 0 then do 

   say "SDSF-Schnittstelle konnte nicht geladen werden" 

   exit 

end 

/* Whatever you do */ 

exit 

5.2. ISFSLASH (Consol-Schnittstelle) 

Mit ISFSLASH wird ein adäquater Ersatz für die TSO-Consolschnittstelle angeboten. In vielen Un-
ternehmen ist deren Nutzung einem sehr eng gesteckten Benutzerkreis vorbehalten, da dort eine 
granulare Nutzung bestimmter Befehle seitens RACF nicht geregelt ist. Somit könnte häufig, der 
Mitarbeiter, welcher die TSO-Colsole aufrufen darf, auch einen SHUTDOWN-Befehl absetzen. Den-
noch wird das Console Command im REXX-Teil beschrieben. 

Unter SDSF dagegen ist im Allgemeinen sehr fein dosiert abgestuft (via RACF), welcher Mitarbeiter 
welches Kommando absetzen darf. Allerdings ist hinsichtlich dieser Definitionen manchmal die Sinn-
frage zu stellen: Warum dürfen unter SDSF manche Dinge nicht von jedermann gemacht werden, 
wenn jedermann mit anderen Mitteln an die gleichen Informationen gelangen kann? Man muss ja 
nur wissen, wie es geht. Security by Obscurity? Lasst euch einfach nicht abwimmeln und fordert bei 
der für das RACF zuständigen Gruppe entsprechende Rechte für die Allgemeinheit. Natürlich habe 
ich aber auch Verständnis dafür, dass RACF-seitig nicht jedem Mitarbeiter jeder Wunsch erfüllt wer-
den kann. 

5.2.1. Variablen für ISFSLASH 

Im Wesentlichen werden nur einige wenige Variablen für ISFSLASH benötigt. Diese sind 

▪ Stemname, welcher die Commands enthält (sofern sie nicht als "Liste von Commands" über-
geben werden, was eher unüblich und wenig änderungsfreundlich ist) 

▪ Stemsize in Form der .0-Variablen, welche die Anzahl der zu berücksichtigenden Stemvari-
ablen enthält 

▪ Stemname, welcher den Output aufnehmen soll. 

5.2.2. ISFSLASH Beispiele 

5.2.2.1. MVS-Befehlsausführung 

Das folgende Beispiel zeigt eine simple Anwendung, in welcher mehrere Befehle ausgeführt werden 
und das Gesamtergebnis in einem Ausgabestem abgelegt wird: 

/* REXX  

*/ 

rc=isfcalls("ON") 

Command.0=3 

Command.1="D R,L" 

Command.2="D T" 

Command.3="D A,L" 

Address SDSF "ISFSLASH (Command.) (WAIT)" 

/* Ausgabe Fehlermeldungen, woweit vorhanden */ 

Say "isfmsg is:" isfmsg 

Say "isfmsg2.0 is:" isfmsg2.0 

if isfmsg2.0 > 0 then do 

   do §=1 to isfmsg2.0 

      Say "ISFMSG2."§ ">>" isfmsg2.§ 

   End 

end 

rc=isfcalls("OFF") 

exit 

5.2.2.2. Befehlsausführung und -Interpretation 

Die Abarbeitung der einzelnen Befehlsergebnisse wäre natürlich auf eine Art und Weise wünschens-
wert, in welcher die Kernaussage der einzelnen Befehle geliefert werden. Das folgende Beispiel 
zeigt die Auswertung der Befehlsergebnisse anhand der Message-IDs: 

/* REXX  

*/ 

rc=isfcalls('ON') 

 

cmd.0=3        /* Number of commands in stem */ 

cmd.1='d r,l'  /* Returned as IEE136I        */ 

cmd.2='d t'    /* Returned as IEE112I        */ 

cmd.3='d a,l'  /* Returned as CNZ4105I       */ 

 

isfdelay=1     /* Set delay interval to 1 second */ 

isfconmod="on" 

isfcmdlim=10 

 

Address SDSF "ISFSLASH (cmd.) (WAIT)" 

do §=1 to isfulog.0 

   select 

      when wordpos("IEE136I",isfulog.§)  > 0 then call CHECK_D_T  

      when wordpos("IEE112I",isfulog.§)  > 0 then call CHECK_D_R_L 

      when wordpos("CNZ4105I",isfulog.§) > 0 then call CHECK_D_A_L 

      otherwise nop  

   end 

end 

 

rc=isfcalls('OFF') 

Exit 0 

 

check_D_T:  

   parse var isfulog.§ . "IEE136I" . "TIME=" Time . "DATE=" Date .  

   parse var date 1 cc 3 yy "." ddd  

   Datum = translate("Tt.Mm.CcJj",date("S",yy !! ddd,"J"),"CcJjMmTt")  

   say "Heute ist der" datum". Es ist jetzt" time;say  

return 

 

CHECK_D_R_L: 

   lpar.  = 0 

   mvesse = "" 

   do §=§ until wordpos("CNZ4105I",isfulog.§) > 0 

      parse var isfulog.§ . . MVS reply . 

      parse var reply 1 pre 2 rnum 

      if pos(pre,"§*") > 0 & datatype(rnum,"W") then do 

         if wordpos(mvs,MVESSE) = 0 then MVESSE = MVESSE mvs 

         lpar.mvs = lpar.mvs + 1 

      end 

   end 

   mvesse = §sortwrd(MVESSE)  /* User Defined Function */ 

   §=§-1 

   say "Offene Meldungen im" #plex()"-Plex:" 

   do $=1 to words(mvesse) 

      sys = word(mvesse,$) 

      say sys !! right(lpar.sys,6,".") 

   end 

return 

 

check_D_A_L: 

   do §=§ to isfulog.0 

      if wordpos("A023034",isfulog.§) > 0 then do 

         say;say "Heinzilein ist auch aktiv!" 

         §=isfulog.0 

         Leave 

      End 

   End 

return 

Es geht natürlich auch etwas komfortabler. Da in der Regel meist nur EIN Command zu einer Zeit 
abgesetzt werden soll, da anderenfalls die Abarbeitung der Rückmeldungen chaotisch werden, bie-
tet sich an, für die Ausführung eines Consol-Commands eine Zentrale REXX zu haben, bei welcher 
der Name des Ergebnis-Stems mitgegeben werden kann. Auf diese Weise kann eine REXX mehrere 
zOS- oder JES2-Commands absetzen, ohne dass sich der Ergebnis-Stem immer überschreibt. 
Wahlweise kann das Command-Ergebnis auch über den Stack zurückgegeben werden. 

Dafür, dass das Ergebnis in einem wahlfreien Stem abgelegt wird, muss das Programm als Function 
gerufen werden. Das RESULT kann dann per Interpret-Befehl umgesetzt werden. 

5.2.2.3. System Symbols 

Ermittlung und Anzeige der definierten System Symbols und deren Werten. 

/* REXX 

   Programmbeschreibung und Update History am Ende 

*/ 

parse source . . RxName . 

call msg "OFF" 

"Newstack" 

Call isfcalls "ON" 

cmd.0=1 

cmd.1="D SYMBOLS" 

isfdelay=1     /* Set delay interval to 1 second */ 

isfconmod="on" 

isfcmdlim=10 

Address SDSF "ISFSLASH (cmd.) (WAIT)" 

call isfcalls "OFF" 

do §=1 to isfulog.0 

   Sym = Strip(space(isfulog.§,1)) 

   if left(Sym,1) = "&" & word(Sym,2) = "=" then do 

      parse var Sym var "=" wert 

      var=strip(var) 

      wert = strip(strip(wert),,'"') 

      queue left(var,10) ">>" wert 

   end 

end 

Queue 

ErgDsn      = userid()".SYMBOLS.DER."mvs 

if sysdsn("'"ErgDSN"'")<>"DATASET NOT FOUND" then "del '"ErgDSN"'" 

"alloc dd(TMP) da('"ErgDSN"') new dsorg(PS)" , 

   "lrecl(70) blksize(70) recfm(F) space(5) tracks" 

"execio * diskw TMP (finis" 

Address ispexec "view dataset('&ErgDsn')" 

"del '"ErgDSN"'" 

"delstack" 

Exit 

 

/* Programmbeschreibung *********************************************** 

 

   Ermittelt die System-Symbols einer LPAR und zeigt sie mit 

   ihren Inhalten in einem TEMP-Dataset. 

 

-- Update History ----------------------------------------------------- 

 

   05.08.2019 Wittemann – Erstellung 

*/ 

 

5.3. ISFLOG (Zugriff auf das zOS-LOG) 

Die Ansteuerung des LOGs sei hier in zwei kleinen Beispielen erläutert. Die Programme sprechen 
für sich selbst. 

/* REXX 

   Work with CURRENT syslog 

*/ 

If isfcalls('ON') <> 0 then exit 

isflinelim=100000 

ADDRESS SDSF "ISFLOG READ TYPE(SYSLOG)" 

suche = "Nudelsuppe"  /* Suchstring */ 

 

SAY 'isfmsg2.0 = 'isfmsg2.0 

SAY 'isfline.0 = 'isfline.0 

 

do §=1 to isfmsg2.0 

   say isfmsg2.§ 

end 

 

do §=1 to isfline.0 /* Process the returned variables */ 

   /* say isfline.§ >> wenn jede Zeile des Logs angezeigt werden sollte*/ 

   IF POS(suche,isfline.§) <> 0 THEN SAY isfline.§ 

End 

rc=isfcalls('OFF') 

Nachfolgend das gleiche Beispiel unter Berücksichtigung eines bestimmten Zeitfensters: 

/* REXX 

   work with 24h Syslog 

*/ 

  

suche = "Blumentopf"    /* Suchstring */ 

if isfcalls('ON') <> 0 then exit 

 

isfsysid="MVS1"                       /* LPAR                   */ 

isfdate="mmddyyyy /"                  /* Datums-Format          */ 

currday=date("C") 

currday=currday-1                     /* Gestern                */ 

isflogstartdate=date("U",currday,"C") /* yesterday in mm/dd/yy  */ 

isflogstarttime=time("N")             /* Aktuelle Zeit          */ 

isflogstopdate=date("U")              /* Tagesdatum in mm/dd/yy */ 

isflogstoptime=time("N")              /* Aktuelle Zeit          */ 

isflinelim=10000                      /* Maximale Anzahl Zeilen */ 

ADDRESS SDSF "ISFLOG READ TYPE(SYSLOG)" 

 

do §=1 to isfmsg2.0 

   say isfmsg2.§ 

end 

 

do §=1 to isfline.0 /* Verarbeitung zurückgegebener Variablen */ 

   IF POS(suche,isfline.§) <> 0 THEN SAY isfline.§ 

End 

rc=isfcalls('OFF') 

exit  

5.4. ISFEXEC (Befehlsausführung) 

ISFEXEC erlaubt die Ausführung von Commands in SDSF. In der Regel erfolgt darüber der Aufruf 
eines Panels. Abhängig davon, welches Panels aufgerufen wird, werden eine Reihe von Stemvari-
ablen gefüllt, welche im Wesentlichen den Spaltenüberschriften des jeweiligen Panels entsprechen.  

Generell wollen wir zwischen den eher einfachen Panels wie PARM, PROC, SE, INIT und derglei-
chen und den komplexeren Panels wie DA, ST, H unterscheiden. Bei den einfacheren kann in aller 
Regel ein oder eine Reihe von parallel gefüllten Stems per Iteration die Ergebnisverarbeitung erfol-
gen. In der jeweiligen .0-Variablen zum Stem liegt die Anzahl der erzeugten Stemvariablen. Dieser 
Wert ist im Grunde für die .0-Variablen aller Stems gleich. 

5.5. ISFACT (Absetzen von Linecommands) 

Über ISFACT können Linecommands abgesetzt werden. Die gängigsten Linecommands in den 
komplexeren Panels (DA, H und ST) sind zweifellos das Fragezeichen (aufklappen des JobLogs), 
die Selektion eines Eintrags (Zeig mir den Eintrag an) und XDC (Druck mir einen Eintrag aus, wobei 
DRUCKEN nicht s ernst gemeint ist). 

Für den Zugriff auf einen einzelnen Eintrag der Ergebnistabelle (des aufgerufenen Panels) ist ein 
eindeutiges Identifizierungsmerkmal notwendig, das sogenannte Token. Der Token-Stem wird pa-
rallel zu den sonstigen Stems der jeweiligen Abfrage gefüllt. Ging beispielsweise eine ST-Abfrage 
voraus, wird für jedes Inkrement der Stems auch ein TOKEN.x gefüllt. der Zugriff auf ein Element 
würde dementsprechend so aussehen (können): 

"ISFACT ST TOKEN('"jobtoken"') PARM(NP ?)(prefix ?_" 

Über NP wird das Feld Linecommand angesprochen. Habt ihr schon mal bemerkt, dass es in den 
SFSF-Panels eine gleichnamige Überschrift hat? Ein "NP ?" ist also gleichbedeutend mit der Bedie-
nereingabe ? als Linecommand im jeweiligen Panel. Das Token stellt die Verbindung zum ge-
wünschten Element her. Die Prefix-Angabe ist optional und hätte, sofern sie kodiert wird, folgende 
Bewandtnis: 

Werden mehrere Abfragen auf SDSF-Elemente durchgeführt, welche die gleichen Stems befüllen 
und soll verhindert werden, dass diese sich gegenseitig überschreiben, wird die Prefix-Angabe vor 
die entsprechenden Variablennamen gesetzt und so eine Eindeutigkeit hergestellt. Der Stepname, 
welcher in der Regel über den Stem STEPN. zurückgegeben wird, läge bezogen auf das Beispiel 
jetzt als ?_STEPN. vor. 

5.6. ISFBROWSE (Lesen eines Eintrags in SDSF) 

ISFBROWSE ist für die "Anzeige" eines SDSF-Elements zuständig, wobei "Anzeige (BROWSE) hier 
nicht das richtige Wort ist. In der ISPF-Oberfläche von SDSF würde sich letztlich BROWSE so zei-
gen, wie man es von BROWSE erwartet. Über die REXX_API führt der Aufruf von Browse lediglich 
zur Befüllung eines Stems, in dem die anzuzeigenden Daten liegen. Sehen wir uns auch dazu ein 
Beispiel an: 

: 

isflinelim     = 10000    /* max Zeilenazahl pro Job f. ISFBROWSE */ 

isfbrowse_parm = ' (JCL) '/* Was soll Browse liefern ?            */ 

outmemb        = 'JCL'SUBSTR(jobid,4) /* Output Member            */ 

tempdsn        = SdsfLib'('outmemb')' /* Output Dataset           */ 

drop isfline. /Prophylaxe, wird von Browse gefüllt */ 

Address SDSF "ISFBROWSE ST TOKEN('"jobtoken"') "isfbrowse_parm 

: 

Und was machen wir jetzt mit dem Ergebnis? Was immer Ihr wollt. Hier ein denkbarer Weg: 

: 

AloParm = "space(10,10) cyl dsorg(ps) lrecl(137) recfm(v) rtddn(dd)" 

call bpxwdyn "alloc" AloParm 

address tso "EXECIO * DISKW" dd "(STEM isfline. FINIS)" 

call edivibro dd /* Zeigt das Ergebnis im VIEW */ 

call bpxwdyn "free dd("ddn")" 

: 

5.7. REXX-Variablen für umfangreichere Panels (DA, ST, H) 

Wir betrachten zunächst einige Variablen, welche in REXX vor Aufruf von umfangreicheren Panels 
(DA, ST, H) gesetzt werden können. Diese Variablen wollen wir in 4 Gruppen einteilen 

5.7.1. Variablen für Filter Commands 

Filtering ist in den meisten Fällen die bessere Art selektiv auf SDSF-Daten zuzugreifen. Werden in 
einer Anwendung aber Filter gesetzt und mehrere Befehle an das SDSF abgesetzt, kann es sein, 
dass die entsprechenden Variablen zurückgesetzt werden müssen. Natürlich kann das mit REXX-
Mitteln geschehen, aber eine simple Art dies zu tun, ist der ISFRESET-Befehl (ist im Command-
Referenz-Teil beschrieben) 

5.7.1.1. ISFPREFIX 

Entspricht der Prefix-Angabe in native SDSF. In der Praxis ist in aller Regel die Anwendung von 
Filtern der bessere Weg. Mit Filtering ist es ratsam, ISFPREFIX="**" zu setzen. 

5.7.1.2. ISFOWNER 

Entspricht der Owner-Angabe in native SDSF. Auch hier drängt sich meist die Nutzung von Filtern 
auf. Mit Filtering ist es ratsam, ISFOWNER="*" zu setzen. 

5.7.1.3. ISFDEST 

Entspricht dem DEST-Befehl In native SDSF und ist gedacht, für ein Filtering abhängig von JES- 
Local Printers. Auf diese Möglichkeit wird nicht weiter eingegangen, da dies heute nicht mehr zeit-
gemäß ist. 

5.7.1.4. ISFFILTER 

5.7.1.5. ISFFILTERMODE 

5.7.1.6. ISFINPUT 

5.7.1.7. ISFSYSNAME 

5.7.2. Variablen für SDSF Commands 

Diese können seit zOS 2.3 vernachlässigt werden. 

5.7.3. Variablen für OPTION Commands 

5.7.3.1. ISFACTIONS 

Ermöglicht, die Befehle für ein bestimmtes Panel in Erfahrung zu bringen, welche über den Stem 
ISFRESP. Zurückgemeldet werden. Das folgende Beispiel zeigt diese Möglichkeit in Verbindung mit 
dem Aufruf des Status-Panels. 

/* REXX  

*/ 

rc=isfcalls('ON') 

isfactions="ON" 

Address SDSF "ISFEXEC ST" 

if rc >0 then exit rc 

Say "Mögliche Actions für das ST Panel sind:" 

do §=1 to isfresp.0 

   Say ">>" isfresp.§ 

End 

rc=isfcalls('OFF') 

exit 

5.7.3.2. ISFCKLIM 

Benennt das Limit an Instanzen für einen Health-Checker, welcher im CHK Panel angezeigt werden 
kann. 

5.7.3.3. ISFCONMODE 

Regelt den Umgang mit Console Namen. Der Default ist ON. Dies bedeutet, dass SDSF für den Fall 
eines Console-Aufrufs, bei dem festgestellt wird, dass der Default Console Name bereits belegt ist, 
den Default-Namen so modifiziert, dass er zur Laufzeit eindeutig ist (siehe auch unter SET CON-
MOD in der Online-Hilfe von SDSF). 

5.7.3.4. ISFCONS 



Inside z/OS Seite 338 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

Würde für den Fall, dass ISFCONMODE auf OFF gesetzt ist, den Default-Console-Namen liefern. 
Da dies aber in der Regel nicht sinnvoll ist, wird hier nicht weiter darauf eingegangen (siehe auch 
unter SET CONSOLE in der Online-Hilfe von SDSF). 

5.7.3.5. ISFDATE 

Regelt das Datumsformat für den ISFLOG-Befehl (siehe auch unter SET DATE in der Online-Hilfe 
von SDSF). 

5.7.3.6. ISFDELAY 

Steuert das Timeout-Verhalten für Befehlsantworten (siehe auch unter SET DELAY in der Online-
Hilfe von SDSF). 

5.7.3.7. ISFDISPLAY 

Enthält die Filter- und Sortierkriterien, z.B.:  

PREFIX=* DEST=(ALL) OWNER=SPUSER1 SYSNAME= 

(siehe auch unter SET DISPLAY in der Online-Hilfe von SDSF) 

5.7.3.8. ISFDISPLAYMODE 

Legt das Format der gleichnamigen Variablen fest. Der Parameter OFF ist unter REXX nicht verfüg-
bar (siehe auch unter SET CONMOD in der Online-Hilfe von SDSF). 

5.7.3.9. ISFDUPDS 

Steuert, ob doppelte SYSOUT-Dateien angezeigt werden. 

5.7.3.10. ISFINPUT 

Steuert, ob SYSIN-Daten mit angezeigt werden (siehe auch unter INPUT Command in der Online 
Hilfe von SDSF). 

5.7.3.11. ISFSCHARS 

Definiert die generischen Platzhalter (siehe auch unter SET SCHARS Command in der Online Hilfe 
von SDSF). 

5.7.3.12. ISFTIMEOUT 

Zeitdauer für SYSPLEX Daten bis zum Timeout (siehe auch unter SET TIMEOUT Command in der 
Online Hilfe von SDSF). 

5.7.4. Variablen für TRACE Commands 

5.7.4.1. ISFSECTRACE 

Bestimmt die Optionen, wenn ein SDSF Securitytrace eingeschaltet wird 

5.7.4.2. ISFMSG2 

Enthält die Trace-Nachricht, wenn ISFSECTRACE auf ON gesetzt ist. 

5.7.4.3. ISFULOG 

Enthält die Security-Trace-Message, wenn ISFSECTRACE WTP gesetzt ist. 

5.7.4.4. ISFTRACE 

Legt die Trace-Option fest, wenn SDSF Trace aktiviert wird 

5.7.4.5. ISFTRMASK 

Legt eine Trace-Maske fest, wenn SDSF Trace aktiviert wird 

5.8. Anzeige von SDSF-Elementen 

Grundsätzlich können wir hier zwei unterschiedliche Verfahren anwenden. Zum einen, können wir 
mittels EXECIO auf ein SDSF-Element zugreifen, vorausgesetzt, die vorangegangene ISFACTion 
war ein "NP SA". SA allokiert im Hintergrund die SDSF-Einträge dynamisch und alles was wir für 
den EXECIO benötigen, ist der DD-Name. 

Zum anderen können wir den Zugriff auch über ISFBROWSE durchführen. Wir betrachten beide 
Varianten anhand eines kleinen Beispielprogramms und ihr entscheidt (fällt unter "Kreative Freiheit 
des Entwicklers"). 

Ich selbst neige zur Variante 2. Aber es gibt immer Ausnahmen 

5.8.1. Ohne ISFBROWSE 

/* REXX 

*/ 

call isfcalls "ON" 

Address SDSF 

isfprefix = "**" 

isfFilter = "JOBID EQ JOB01477" 

"ISFEXEC H" 

"ISFACT H TOKEN('"TOKEN.1"') PARM(NP SA)" 

do _1=1 to sdsfout.0 

   address tso "EXECIO * DISKR" isfddname._1 "(STEM SDSFOUT. FINIS" 

   do _2=1 to isfddname.0 

      say sdsfout._2 

   end 

end 

exit 

5.8.2. Mit ISFBROWSE 

/* REXX 

*/ 

call isfcalls "ON" 

Address SDSF 

isfprefix = "**" 

isfFilter = "JOBID EQ JOB01477" 

"ISFEXEC H" 

"ISFBROWSE H TOKEN('"TOKEN.1"')" 

do §=1 to isfline.0 

   say isfline.§ 

end 

exit 

5.9. RGEN, oder Warum selber programmieren? 

Der Befehl RGEN scheint auf den ersten Blick eine allglückseeligmachende Lösung für den Umgang 
mit dem REXX-API. Der Befehl kann in jedem beliebigen Panel von SDSF eingegeben werden und 
generiert den Quellcode für eine Abfrage der entsprechenden Paneldaten.  

Gehen wir von folgender Situation aus: Es soll in Erfahrung gebracht werden, welche Scheduling 
Environments definiert und für welche LPARs sie gesetzt sind. 

5.9.1. Abfrage SE (Scheduling Environments), erstellt durch RGEN 

000001 /* REXX */ 

000002 Arg debug 

000003 

==MSG>    Important: Copy this generated exec from temporary dataset 

==MSG>      SYS21137.T084606.RA000.XCG5540.R0124965 

==MSG>    and edit that copy to prevent your changes from being lost. 

==MSG> 

000004 /**************************************************************** 

000005 * 

000006 *  SDSF RGEN Generated EXEC 

000007 * 

000008 *  This exec was generated by the SDSF RGEN command on 

000009 *  Monday 2021/05/17 at 09:05:34.84. 

000010 * 

000011 *  5650-ZOS 

000012 *  SDSF level = z/OS 02.04.00 (HQX77C0) 

000013 * 

======    Use this exec as a starting point for writing your own execs. 

======  

======    The RGEN command generates an exec that accesses the current 

======    panel and shows how to use special variables. 

======  

======    For more information and examples, use the SDSF REXXHELP 

======    command.  To search SDSF's help, use SEARCH search-string. 

======  

000014 *  Operation =  

000015 * 

000016 *    - Access primary panel SE 

000017 * 

000018 *****************************************************************/ 

000019  

=NOTE= Tip: All SDSF/REXX execs must include the following statement: 

000020 rc=isfcalls('ON') 

000021 

000022 trace o 

000023  

=NOTE= Tip: The verbose option provides additional diagnostics 

=NOTE=      when invoking SDSF services. 

000024 if debug<>"" then   /* If debug mode */ 

000025   verbose="VERBOSE"  /* .. use SDSF verbose mode */ 

000026 else 

000027   verbose="" 

000028  

000029 /*----------------------------------------------*/ 

000030 /* Configure environment with special variables */ 

000031 /*----------------------------------------------*/ 

=NOTE= Tip: You must be authorized to the corresponding command 

=NOTE=      to set the variable. 

=NOTE=  

=NOTE= Tip: Not all variables may be needed in your exec. 

=NOTE=  

000032 isfprefix='XCG5540*'    /* Corresponds to PREFIX command */ 

000033 isfowner='*'      /* Corresponds to OWNER command */ 

000034 isfsysname=''   /* Corresponds to SYSNAME command */ 

000035  

000036 isfdest=' ' !! ,    /* Dest name 1 */ 

000037         ' ' !! ,    /* Dest name 2 */ 

000038         ' ' !! ,    /* Dest name 3 */ 

000039         ' '         /* Dest name 4 */ 

000040  

000041  

000042 /* Access the SE panel */ 

000043 Address SDSF "ISFEXEC 'SE' (" verbose ")" 

000044 lrc=rc 

000045  

=NOTE= Tip: Always check the return code after each request. 

000046 call msgrtn  "ISFEXEC 'SE'"   /* List messages */ 

000047 if lrc<>0 then   /* If command failed */ 

000048   do 

000049     Say "** ISFEXEC failed with rc="lrc"." 

000050     exit 20 

000051   end 

000052  

=NOTE= Tip: The special variable sdsfocols is a word delimited 

=NOTE=      list of column names returned on the request. 

000053 call colsrtn isfrows "." sdsfocols /* List all rows and columns */ 

000054  

000055  

=NOTE= Tip: All SDSF/REXX execs must finish with the following statement:  

000056 rc=isfcalls('OFF') 

000057  

000058 Exit 0 

000059  

000060  

000061 /***************************************************************** 

000062 * 

000063 * NAME =  

000064 *   msgrtn 

000065 * 

000066 * FUNCTION = 

000067 *   List all messages in the isfmsg and isfmsg2. Variables 

000068 * 

000069 * INPUT =  

000070 *   req - Request being processed 

000071 * 

000072 * EXPOSED VARIABLES =  

000073 *   isfmsg   - Short message 

000074 *   isfmsg2. - Numbered messages 

000075 * 

000076 * OUTPUT =   

000077 *   Messages written to terminal 

000078 * 

000079 ****************************************************************/ 

000080 msgrtn: Procedure expose isfmsg isfmsg2. 

000081 Arg req 

000082  

000083 /*---------------------------*/ 

000084 /* Process numbered messages */ 

000085 /*---------------------------*/ 

000086 Say "** Numbered messages associated with" req "follow ..." 

=NOTE=  

=NOTE= Tip: The isfmsg2. stem contains numbered messages 

=NOTE=      associated with the request and isfmsg2.0 contains 

=NOTE=      a count of the number of variables that follow. 

000087 do ix=1 to isfmsg2.0 

000088   Say isfmsg2.ix 

000089 end 

000090  

000091 if isfmsg<>"" then    /* If short message present */ 

000092   do 

000093     Say "** Short message associated with the request is:" isfmsg 

000094   end 

000095  

000096 return 

000097  

000098  

000099 /***************************************************************** 

000100 * 

000101 * NAME = 

000102 *   colsrtn 

000103 * 

000104 * FUNCTION = 

000105 *   List all rows and their column values 

000106 * 

000107 * INPUT =  

000108 *   numrows - number of rows to process 

000109 *   pfx     - column variable prefix or "." if none 

000110 *   ocols   - word delimited column names to process 

000111 * 

000112 * EXPOSED VARIABLES =  

000113 *   None 

000114 *  

000115 * OUTPUT =  

000116 *   Responses written to terminal 

000117 *  

000118 *****************************************************************/ 

000119 colsrtn: 

000120 Arg numrows pfx ocols 

000121 Say "Number of rows to process: " numrows 

000122  

000123 do rowix=1 to numrows  /* Loop for all rows */ 

000124   Say "Now processing row" rowix "..." 

000125  

000126   do colix=1 to words(ocols)  /* Loop for all columns */ 

000127  

000128     if pfx="." then   /* If no prefix */ 

000129       pfx="" 

000130  

000131     varname=pfx!!word(ocols,colix)!!'.'!!rowix 

000132  

000133     Say "  Column" varname '=' value(varname) 

000134   end     /* For all columns */ 

000135 end   /* For all rows */ 

000136  

000137 return 

Wenn die special Lines mitgezählt werden, sind das 165 Lines of Code. Wirft man diese per RESET 
alle raus und löscht die größeren Kommentare, verbleiben immer noch über 140 Zielen. Ich möchte 
aber ausdrücklich darauf hinweisen, dass ich Kommentare in einem Programm für sehr sinnvoll 
halte. 

Brauchen wir dieses ganze Coding? Nein! Es ist vielleicht eine Hilfe, wenn man die ersten Gehver-
suche mit REXX_SDSF macht, aber was muss man den schon groß wissen? Im Wesentlichen 
würde erst einmal reichen, welche Variablen die benötigten Informationen enthalten. Ein Programm, 
welches das liefert ist im weiteren Verlauf zu sehen (PANFLDS). Ansonsten reicht ein Minipro-
gramm, das die gewünschte Information liefert: 

/* REXX 

*/ 

Call isfcalls "ON" 

Address sdsf "isfexec se" 

Say isfcols 

Exit 

Und schon wissen wir, welche Stems wir verarbeiten wollen. Dementsprechend wäre das entspre-
chende Programm dazu überschaubar: 

5.9.2. Abfrage SE (Scheduling Environments), selbst kodiert 

/* REXX 

*/ 

call isfcalls "ON" 

address SDSF "isfexec SE" 

say isfcols 

do §=1 to schenv.0 

   say left(schenv.§,10,".") systems.§ 

end 

exit 

Kein Kommentar – entscheidt selbst. 

5.10. Anwendungsbeispiele 

Die nachfolgenden Seiten zeigen einige einfache Anwendungsbeispiele. Natürlich kann man alle 
diese Beispielprogramme auch deutlich komfortabler (auch komplexer) gestalten. Vieles würde man 

in der täglichen Praxis ausführlicher kodieren. Fürs Erste geht es aber nur darum die Möglichkeiten 
zu aufzuzeigen. 

5.10.1. PANFLDS, Welche Variablen liefert welches Panel 

Das Beispiel zeigt ein Programm, über welches man ermitteln kann, welche Felder (Variablen oder 
Stems) ein Panel zurückliefern würde, wenn es via REXX gerufen wird. Man könnte das optisch über 
die ISPF-Services aufpeppen und natürlich wäre im Vorfeld gegebenenfalls über eine Anfrage bei 
RACF sinnvoll, ob der Aufrufer das entsprechende Panel überhaupt sehen darf. 

/* REXX * PANFLDS 

   List all REXX variables/columns by SDSF Panel 

*/ 

Arg IsfPan 

if isfcalls('ON') <> 0 then call abend "SDSF ist nicht verfügbar" 

address SDSF "ISFEXEC" IsfPan "(ALTERNATE DELAYED)" 

if RC > 0 then call abend "Panel" IsfPan "nicht verfügbar" 

AllCols = isfcols  isfdcols 

do §=1 to words(allcols) 

   say right(§,4) ">>" word(allcols,§) 

end 

exit 

 

abend: zedlmsg=arg(1);address ispexec "setmsg msg(isrz000)";exit 

5.10.2. DSNPROC, Dateinamen der JES-Proclib-Verkettung 

Häufig sieht man Programme, welche die Dateien der JES-Proclib-Verkettung untersuchen, wobei 
die Namen der Dateien oft hardcoded im Programm stehen. Warum nicht die entsprechenden Da-
tasets zur Laufzeit ermitteln? Das kann auf jeder LPAR, auf jedem Plex benutzt werden und ist 
nebenbei auch immer richtig ;-) 

/* REXX * Ermittlung der PROCLIB-Verkettung eines bestimmten DD-Namens 

*/ 

parse source . RxTyp RxName . 

arg parm . 

 

main_start: 

   if isfcalls("ON") <> 0 then call abend "Kein SDSF da." 

   call set_vars 

   call get_proclib_info 

main_ende: 

if RxTyp <> "COMMAND" then return DsListe 

call Show_Result 

exit 

 

get_proclib_info: 

   isfcols = "DSNAME DDNAME SEQ" /* Nur diese Felder werden gebraucht */ 

   address sdsf "isfexec PROC" 

   DsListe = 

   SeqList = 

   do §=1 to dsname.0 

      if ddname.§ <> dd.PARM then iterate  

      DsListe = DsListe dsname.§ 

      SeqList = SeqList Seq.§ 

   End 

return 

 

set_vars: 

   if wordpos(parm,"JCL LGN STC") = 0 then parm = "JCL" 

   type.JCL = "Cataloged Procedures" 

   type.LGN = "Logon Routines" 

   type.STC = "JCL for Started Tasks" 

   dd.JCL   = "PROC00" 

   dd.LGN   = "PROC01" 

   dd.STC   = "PROC02" 

return 

 

Show_Result: 

   say left(type.Parm,60,"_");say 

   say "SEQ" left("Dateiname",44,"_")  

   do §=1 to words(DSListe) 

      say right(word(SeqList,§),3) word(DsListe,§) 

   end 

return 

 

abend: procedure;zedlmsg=arg(1);address ispexec "setmsg msg(isrz000)";exit 

Auf die gleiche Art könnte auch die Parmlib-Verkettung ermittelt werden. Mit ein wenig Phantasie 
würde das EIN Programm erledigen können, das eine etwas ausgiebigere Parameterübergabe be-
nötigen würde. 

5.10.3. SDSFSEND, Send Message an alle TSO-User 

/*REXX 

   Führt im SDSF einen ST Command aus und ermittelt alle aktiven 

   LOGON-Sessions um einen SEND-Befehl an die User abzusetzen 

*/ 

 

text = 'Achtung: auf' mvsvar("SYSNAME") "erfolgt in 10 Minuten IPL !" 

if isfcalls("ON")<> 0 then call abend "ISFCALLS ended RC="rc 

isffilter = "JTYPE EQ TSU QUEUE EQ EXECUTION" 

Address SDSF "ISFEXEC ST" 

DO §=1 TO jname.0 

   ADDRESS TSO "SEND '"text"' U("strip(jname.§)")" 

END 

rc=isfcalls("OFF") 

exit 

 

abend: parse arg zedlmsg;address ispexec "setmsg msg(ISRZ000)";exit 

5.10.4. SDSFSRCH, Suche Step/DDname in Joblog 

Diese kleine Routine zeigt, wie man zur Laufzeit oder auch danach ein bestimmtes DD-Statement 
eines bestimmten Steps lesen kann. Da ja beim Zugriff auf ein Element im SDSF intern eine Zuwei-
sung auf ein temporäres File für die Anzeige erfolgen muss, liegt das "Geheimnis" lediglich im Wis-
sen um den entsprechenden DD-Namen, über den simpel mit EXECIO gelesen werden kann. 

/* REXX 

   SDSFCHK * Prüfung eines Joblogs 

*/ 

parse source . RxTyp RxName . 

 

call set_vars 

 

call isfcalls "ON" 

Address SDSF 

"ISFEXEC ST" 

if RC <> 0 then call abend "RC:" rc "bei ISFEXEC ST" 

 

"ISFACT ST TOKEN('"TOKEN.1"') PARM(NP ?)" 

do _1=1 to stepn.0 

   if Stepn._1 <> StepName ! DDName._1 <> DDName then iterate 

   "ISFACT ST TOKEN('"TOKEN._1"') PARM(NP SA)" 

   address tso "EXECIO * DISKR" isfddname.1 "(STEM SDSFOUT. FINIS" 

   do _2=1 to SDSFOUT.0 

      SdsfOut = substr(sdsfout._2,2) 

      if wordpos(word(SDSFOUT,1),MsgIDs) > 0 then do 

         interpret "CALL" word(SDSFOUT,1) 

         say result 

      end 

   end 

end 

exit 

 

set_vars: 

   if sysvar("SYSENV") = "FORE" then do 

      /* Im Foreground muss die Jobnummer vorgegeben werden */ 

      jnum  = "JOB03308" 

   end 

   else do 

      /* Im Batch wird sie ermittelt. Meist als UPRO verfügbar */ 

      tcb      = storage(21c,4) 

      jscb     = storage(d2x(c2d(tcb)+180),4) 

      ssib     = storage(d2x(c2d(jscb)+316),4) 

      jobnum   = strip(storage(d2x(c2d(ssib)+15),5),l,0) 

      jnum = "JOB"jobnum 

   end 

 

   StepName = "IEBCOPY" 

   DDName   = "SYSPRINT" 

   isfprefix = "**" 

   isfFilter = "JOBID EQ" JNum 

 

   MsgIds = "" , 

            "IEB1135I"  ,    /* Version von IEBCOPY   */ 

            "IEB1013I"  ,    /* Eingabedatei          */ 

            "IEB1014I"  ,    /* Ausgabedatei          */ 

            "IGW01551I" ,    /* Member kopiert        */ 

            "IGW01550I" ,    /* x von y Members       */ 

            "IEB177I"   ,    /* Problemmember         */ 

            "IEB147I"   ,    /* Cond Code             */ 

            !! "" 

Return 

 

IEB1135I: 

   parse var SdsfOut . "LEVEL" lvl "DATED" dat . 

return "Start IEBCOPY  Version" strip(lvl) "vom" , 

       translate("Tt.Mm.CcJj",dat,"CcJjMmTt") 

 

IEB1013I: 

   parse var SdsfOut . "DSN=" DsIn . 

return "Eingabedatei:" DsIn 

 

IEB1014I: 

   parse var SdsfOut . "DSN=" DsOut . 

return "Ausgabedatei:" DsOut 

 

IGW01551I: 

return left(word(SdsfOut,3),8) "wurde kopiert" 

 

IGW01550I: 

   parse var SdsfOut . anz "OF" von . 

return strip(anz) "von" von "Members wurden kopiert" 

 

IEB177I: 

return left(word(SdsfOut,2),8) "ausgewählt aber nicht gefunden" 

 

IEB147I: 

return "Verarbeitung endet mit Cond Code" word(SdsfOut,6) 

 

Abend: 

   parse arg zedlmsg 

   if sysvar("SYSENV") = "BACK" then say zedlmsg 

   else address ispexec "setmsg msg(isrz000)" 

   if isfmsg <> "" then Say RxName "ISFMSG >>" isfmsg 

   do §=1 to isfmsg2.0 

      Say RxName "ISFMSG2."§ ">>" isfmsg2.§ 

   End 

Exit 

Jetzt unterstellen wir folgende JCL, die IEBCOPY benutzt. Die ADR*-Members sind vorhanden. 
MURKS und PFUSCH dagegen nicht. 

//IEBCOPY  EXEC PGM=IEBCOPY 

//IN       DD DISP=SHR,DSN=SP1.EDU.DATA 

//OUT      DD SPACE=(CYL,(10,10,20)) 

//SYSPRINT DD SYSOUT=* 

//SYSIN    DD * 

  COPY INDD=((IN,R)),OUTDD=OUT 

  SELECT MEMBER=ADR1 

  SELECT MEMBER=ADR2 

  SELECT MEMBER=ADR3 

  SELECT MEMBER=MURKS 

  SELECT MEMBER=PFUSCH 

/* 

//CHECKLOG EXEC PGM=IKJEFT01,PARM=SDSFCHK           Start-Rexx 

//SYSEXEC  DD   DISP=SHR,DSN=SP1.TOOLS.EXEC         Enthält das SDSFCHK 

//SYSTSIN  DD DUMMY 

//SYSTSPRT DD SYSOUT=* 

Die Überprüfung der Verarbeitung erfolgt in einem angehängten Step über die REXX SDSFCHK 

Hier das Original-LOG (SYSPRINT des IEBCOPY): 

IEB1135I IEBCOPYA FMID HDZ2240  SERVICE LEVEL UJ00519  DATED 20190816... 

IEB1035I XCG55409  IEBCOPYA 17:38:58 THU 22 APR 2021 PARM='' 

  COPY INDD=((IN,R)),OUTDD=OUT 

  SELECT MEMBER=ADR1 

  SELECT MEMBER=ADR2 

  SELECT MEMBER=ADR3 

  SELECT MEMBER=MURKS 

  SELECT MEMBER=PFUSCH 

IEB1013I COPYING FROM PDSE  INDD=IN       VOL=ET0014 DSN=SP1.EDU.DATA 

IEB1014I                   OUTDD=OUT      VOL=EV1017 

DSN=SYS21112.T173858.RA000. 

IGW01551I MEMBER ADR1  HAS BEEN COPIED 

IGW01551I MEMBER ADR2  HAS BEEN COPIED 

IGW01551I MEMBER ADR3  HAS BEEN COPIED 

IGW01550I 3 OF 5 SPECIFIED  MEMBERS WERE COPIED 

IEB177I MURKS    WAS SELECTED BUT NOT FOUND IN ANY INPUT DATA SET 

IEB177I PFUSCH   WAS SELECTED BUT NOT FOUND IN ANY INPUT DATA SET 

IEB151I JOB HAS TERMINATED WITH ERROR(S) 

IEB147I END OF JOB - 4 WAS HIGHEST SEVERITY CODE 

Und zu dieser Erkenntnis kommt das Checker-Programm: 

Start IEBCOPY  Version UJ00519 vom 16.08.2019 

Eingabedatei: SP1.EDU.DATA 

Ausgabedatei: SYS21112.T173858.RA000.XCG55409.R0238719 

ADR1     wurde kopiert 

ADR2     wurde kopiert 

ADR3     wurde kopiert 

3 von 5 Members wurden kopiert 

MURKS    ausgewählt aber nicht gefunden 

PFUSCH   ausgewählt aber nicht gefunden 

Verarbeitung endet mit Cond Code 4 

Wenn man weiß, wonach man sucht (Fehler, Hinweise, sonstiges), kann man alles automatisieren. 
Und wer liest schon gerne JOBLOGs am laufenden Band?  

Auch hier könnte man natürlich vieles noch mehr standardisieren. Ein Programm in dieser Art gehört 
in jede Toolbox. Ich habe eins, das per Parameter gesteuert wird und seine Ergebnisse als interpre-
tierbaren String (bilden einen Stem) oder im Stack oder auch in einer Datei ablegt. 

5.10.5. SDSFSRCH, die "alte" Art (als abschreckendes Beispiel) 

Bitte nicht zu kritisch betrachten. So war es eben früher. Und so ist es leider in vielen Fällen auch 
noch heute, weil es ja in zOS nie etwas Neues gibt. Ist ja ein veraltetes System das keiner mehr 
braucht (ich kann den Unsinn nicht mehr hören. Auch unter zOS könnte man dazulernen, sofern 
man wollte). 

Ich habe das irgendwo gefunden und will es niemandem vorenthalten. Bitte nicht nachmachen. 

/* REXX */ 

Arg jname jnum ddsearch dsnpref 

if jname = "" ! ^datatype(jnum,"W") then exit 12 

call set_vars 

call sdsf1 

 

"EXECIO * DISKR ISFOUT (STEM SDSF. FINIS" 

sstring = "" !! , 

   'SDSF JOB DATA SET DISPLAY - JOB' Left(jname,8) '(JOB'jnum')' 

lstring = Length(sstring) 

Do i = 1 To sdsf.0 

   If Substr(sdsf.i,3,lstring) = sstring Then do 

      found = true 

      Iterate 

   End 

   If found Then do 

      dsid = Substr(sdsf.i, 36, 3) 

      If Datatype(dsid,'NUM') = true Then do 

         ddname = Substr(sdsf.i, 8 , 8) 

         if ddsearch <> "" then do 

            if ddname=ddsearch then Call sdsf2 jname jnum dsid ddname 

         end 

         else do 

            Call sdsf2 jname jnum dsid ddname 

         End 

      End 

   End 

End 

 

Exit 

 

sdsf1: procedure expose jname jnum 

   "ALLOC F(ISFOUT) NEW DELETE REU space(10,10) tracks" , 

      "LRECL(133) RECFM(F,B,A) DSORG(PS) RELEASE" 

   "ALLOC F(ISFIN) TRACKS SPACE(1) REUSE" 

   Queue "PRE" jname 

   Queue "ST" 

   Queue "FILTER JOBID EQ JOB"jnum 

   Queue "F" jname 

   Queue "++?" 

   "EXECIO" QUEUED()" DISKW ISFIN (FINIS" 

   Address ISPEXEC "SELECT PGM(ISF) PARM('++99,255')" 

Return 

 

sdsf2: procedure expose ddsearch dsnpref 

   Arg jobname jobnum dsid ddname 

   odsn = dsnpref".DSID"Right(dsid, 3, 0)"."Strip(ddname) 

   "ALLOC F(ISFOUT) NEW DELETE REU" , 

      "TRACKS SPACE(1,10) LRECL(133) RECFM(F,B,A) DSORG(PS)" 

   "ALLOC F(ISFIN) TRACKS SPACE(1) REUSE" 

   "ALLOC F(SDSFOUT) DA('"odsn"') NEW REUSE TRACKS", 

      "SPACE(10, 10) LRECL(133) RECFM(F B) DSORG(PS) RELEASE" 

   Queue "PRE" jobname 

   Queue "ST" 

   Queue "FILTER JOBID EQ JOB"Right(jobnum, 5, 0) 

   Queue "F" jobname 

   Queue "++?" 

   Queue "FILTER DSID" dsid 

   Queue "F" ddname 

   Queue "++S" 

   Queue "RESET" 

   Queue "PRINT FILE SDSFOUT" 

   Queue "PRINT" 

   Queue "PRINT CLOSE" 

   "execio" Queued()" diskw isfin (finis" 

   Address ISPEXEC "SELECT PGM(ISF) PARM('++99,255')" 

Return 

 

set_vars: 

   jnum  = Right(jnum, 5, 0) 

   true  = 1 

   false = 0 

   found = false 

   if dsnpref = "" then dsnpref = userid()'.SDSFOUT' 

   dsnpref = dsnpref !! ".JOB"jnum 

return 

6. DSNREXX, die Schnittstelle zu DB2 

Hier soll keinesfalls eine detaillierte Abhandlung über die Innereien von DB2 und die letzten Finessen 
im Bereich von SQL dargeboten werden. Es geht in erster Linie darum, den Umgang mit DB2 aus 
einem REXX-Programm heraus aufzuzeigen. 

Dafür wird anhand ein paar kleiner Bespiele gezeigt, wie der Umgang mit den Ergebnistabellen von-
statten geht (Cursorsteuerung). 

Da im täglichen Leben DB2 (zur Datenhaltung) und ISPF (zur Visualisierung der Daten) Hand in 
Hand gehen und im Grunde eine untrennbare Einheit bilden, wird über eine kleine Anwendung ge-
zeigt, wie Tablespaces und Tabellen erstellt werden können, Tabellen initial geladen und Daten im 
DB2 über einen ISPF-Dialog gepflegt werden können. 

6.1. SQL? 

In der Praxis werden alle Befehle, welche in DB2 angeboten werden, als SQL (Structured Query 
Language) bezeichnet. Generell wird aber sehr wohl unterschieden, um "welche Art" von SQL es 
sich letztlich handelt: 

6.1.1. DDL (Data Definition Language) 

Hierzu gehören alle Befehle, welche für die Definition von Objekten in einer Datenbank benutzt wer-
den  

Befehl Bedeutung 

CREATE Erzeugt eine Datenbank oder ein darin liegendes Objekt wir Tabelle, View, 
Index und dergleichen  

DROP Löscht ein Objekt aus einer Datenbank 

ALTER Führt Strukturveränderungen durch 

COMMENT Fügt Kommentare in das Data Dictionary ein 

RENAME Umbenennen eines Objekts in einer Datenbank 

6.1.2. DQL (Data Query Language) 

Für die Ausführung von Abfragen 

Befehl Bedeutung 

SELECT Beschaffung von Informationen aus einer Datenbank 

6.1.3. DML (Data Manipulation Language) 

Alles, was mit der Veränderung von Datensätzen zu tun hat. 

Befehl Bedeutung 

INSERT Einfügen eines Datensatzes 

UPDATE Änderung eines bestehenden Satzes 

DELETE Löschen eines Datensatzes 

6.1.4. TCL (Transaction Control Language Language) 

Arbeiten mit Transaktionen zusammen. Unter einer Transaktion verstehen wir einen oder mehrere 
Befehle, welche als Einheit betrachtet werden. Ist einer dieser Befehle fehlerhaft, so gilt die gesamte 
Transaktion (alle dazu gehörenden Befehle) als fehlerhaft. 

Befehl Bedeutung 

COMMIT Bestätigt eine Transaktion, Veränderungen im DB2 sind rechtens. 

ROLLBACK Alle zur Transaktion gehörenden Befehle werden rückgängig gemacht. 

6.2. Programmbeispiele 

Die folgenden Seiten zeigen ein paar Beispiele, wie mit Abfragen und der Ergebnisverarbeitung von 
DB2-Tabellen umgegangen wird. 

Ein ewiges Ärgernis ist, dass die IBM unter TSO keine dynamische Veränderung von STEPLIB er-
laubt. Für einen vernünftigen Umgang mit Programmen unter TSO ist dies aber unabdingbar. Aus 
diesem Grund werden in den Beispielen die Programme LIBADD (zur Erweiterung von Steplib) und 
LIBRESET (für die Rücknahme) aufgerufen. Dahinter verbergen sich zwei kleine REXXe, welche 
das Load Module $TEPLIB aufrufen. $TEPLIB ist ein Assemblerprogramm, welches diese Dynamik 
ermöglicht und kann als Sourcecode kostenlos von WWW.CBTTAPE.ORG heruntergeladen wer-
den. 

6.2.1. Ermittlung der Spalten einer Tabelle 

/* REXX * Demo für den Umgang mit DSNREXX 

   Abfrage der Spalten einer Tabelle 

*/ 

arg TabName Max_Cols 

call msg "OFF" 

 

start_main: 

   call Set_Vars 

   call Activate_Dsnrexx 

   address dsnrexx 

   call Db2_Connect 

   call Get_Columns_of_TableName 

   call Show_Columns_of_TableName TabName"~"Result"~"Max_Cols 

   call Db2_Disconnect 

ende_main: 

exit 

 

set_vars: 

   TabName = space(translate(TabName," ","."),0) /* if not given     */ 

   if TabName = "" then TabName = "SYSPACKAGE"   /* if not given     */ 

   SysTab = left(TabName,3) = "SYS"              /* if System-Table  */ 

return 

 

activate_dsnrexx: 

   /****************************************************************** 

   * Hinzufügen einer DB2-Bibliothek um das DSNREXX-Environment 

   * laden zu können. 

   */ 

http://www.cbttape.org/

   call LIBADD "SYSDB.DB20000.DSNLOAD" 

   address tso "SUBCOM DSNREXX" 

   dbRC = 0 

   if RC then dbRC = RxSubCom('ADD','DSNREXX','DSNREXX') 

   if dbRC <> 0 then call abend "DSNREXX konnte nicht geladen werden." 

return 

 

db2_connect: 

   /****************************************************************** 

   * Die Verbindung zu DB2Z wird hergestellt. Ist dies nicht möglich, 

   * wird das Programm beendet. 

   */ 

   "CONNECT DB2Z" 

   if SQLCODE <> 0 then call abend "Kein Connect zu DB2U möglich." 

Return 

 

Get_Columns_of_TableName: Procedure expose TabName Systab 

   /****************************************************************** 

   * Erst wird das SELECT-Statement für die Abfrage im DB2 in der  

   * Variablen SQLSTMT hinterlegt. Das wäre generell nicht notwendig, 

   * ist aber praktisch, weil bei Auftreten eines Fehlers der  

   * jeweilige Inhalt von SQLSTMT ausgegeben erden kann und auf diese 

   * Weise immer zu sehen ist, was der letzte Befehl war. 

   */ 

   Cols    = 

   sqlstmt ="" !! , 

   "SELECT NAME" , 

      "FROM DB2Z.SYSIBM.SYSCOLUMNS" , 

      "WHERE TBNAME = '"TabName"'" 

   if SysTab then SqlStmt = SqlStmt "AND TBCREATOR = 'SYSIBM'" 

 

   /****************************************************************** 

   * Der DECLARE definiert einen Cursor, über welchen nach dem SELECT 

   * auf die einzelnen Ergebniszeilen der Abfrage zugegriffen werden 

   * Der PREPARE führt den SELECT-Befehl aus 

   * Der OPEN öffnet den Cursor 

   * Der FETCH liefert einen Datensatz. Sind alle Datensätze  

   * abgearbeitet, wird bei erneutem FETCH nach dem letzten Satz 

   * der SQLCODE 100 gesetzt (ist vergleichbar zur EOF-Erkennung) 

   */ 

   "execsql DECLARE C1 CURSOR FOR S1" 

   "execsql PREPARE S1 FROM :SQLSTMT" 

   "execsql OPEN C1" 

   "execsql FETCH C1 INTO :CNAME" 

   do while sqlcode = 0 

     Cols = strip(Cols strip(CNAME)) 

     "execsql FETCH C1 INTO :CNAME" 

   End 

   "execsql CLOSE C1" 

return Cols 

 

Db2_Disconnect: 

   /****************************************************************** 

   * Trennt die Verbindung zu DB2Z und nimmt die dynamische 

   * Erweiterung der Steplib-Definitionen zurück 

   */ 

   "DISCONNECT" 

   call LIBRESET "SYSDB.DB20000.DSNLOAD" 

return 

 

Show_Columns_of_TableName: procedure 

   /****************************************************************** 

   * Die ermmittelten Spaltennamen der DB2-Tabelle, welche in der  

   * Subroutine Get_Columns_of_TableName als Wortkette in der Variablen 

   * COLS gebildet wurden, werden ausgegeben 

   */ 

   arg Table "~" Columns "~" Max 

   if ^Datatype(Max,"W") ! Max > words(Columns) then Max=Words(Columns) 

   if Max = words(Columns) then Txt = "Alle" 

   else                         Txt = "Die ersten" Max 

   say left(Txt "Spalten der Tabelle:" Table" ",70,"-");say 

   do §=1 to Max 

      say right(§,3) ">>" word(columns,§) 

   end 

return 

 

abend: procedure 

   arg zedlmsg 

   address ispexec "setmsg msg(isrz000)" 

exit 12  

6.2.2. Ermittlung der Package-Informationen 

Natürlich würde man das Programm in der Praxis etwas benutzerfreundlicher gestalten. Man könnte 
einen kleinen ISPF-Dialog einbauen, über den der Bediener die Namen der Lade- und DBRM-Bibli-
othek eingeben könnte und man würde auch eine Timestamp-Prüfung in DBRM und Load durchfüh-
ren und das Ergebnis entsprechend anzeigen. Auf diese Weise könnte man frühzeitig einen SQL-
CODE -805 vermeiden (Package nicht gefunden). Wir wollen hier aber nur die REXX-DB2-Schnitt-
stelle betrachten. 

Eine kleine ISPF-REXX-DB2-Anwendung befindet sich am Ende des Kapitels. 

/* REXX * Infos aus der SYSIBM.SYSPACKAGES 

 

*/ 

arg Package _DB2 . 

call msg "OFF" 

 

start_main: 

 

   call set_vars 

   call activate_dsnrexx 

   address dsnrexx 

   call db2_connect 

   call read_packtab_from_db2 

   "DISCONNECT" 

   call LIBRESET "SYSDB.DB20000.DSNLOAD" 

ende_main: 

 

"tbend &packtab" 

 

Exit 

 

set_vars: 

   if Package = "" then Package = "PROGABC"   /* if not given – Test */ 

   if _DB2    = "" then _DB2    = "DB2Z"      /* if not given – Test */ 

Return 

 

activate_dsnrexx: 

   call LIBADD "SYSDB.DB20000.DSNLOAD" 

   address tso "SUBCOM DSNREXX" 

   dbRC = 0 

   if RC then dbRC = RxSubCom('ADD','DSNREXX','DSNREXX') 

   if dbRC <> 0 then call abend "DSNREXX konnte nicht geladen werden." 

Return 

 

db2_connect: 

   "CONNECT" _DB2 

   if SQLCODE <> 0 then do 

      CALL SQLCA 

      call abend "Kein Connect zu" _DB2 "möglich." 

   End 

Return 

 

read_packtab_from_db2: 

   SQLSTMT  = "select name, owner, contoken, timestamp, pdsname" , 

                "from" _DB2".sysibm.syspackage" , 

                "where name like '"package"'" 

   "EXECSQL DECLARE C1 CURSOR FOR S1" 

   "EXECSQL PREPARE S1 FROM :SQLSTMT" 

   "EXECSQL OPEN C1" 

 

   rec_found = 0 

   do until SQLCODE <> 0 

 

      "EXECSQL FETCH C1 INTO" , 

         ":PGM ,"       , /* Programmname        */ 

         ":OWNER ,"     , /* Wer hat's gebunden  */ 

         ":CONTOKEN,"   , /* Identifier          */ 

         ":TIMESTMP ,"  , /* Timestamp           */ 

         ":DSN"           /* DBRM-Library        */ 

      Select 

         when SQLCODE = 0 then do 

            ConToken = c2x(ConToken) 

            ConToken = left(ConToken,8) right(ConToken,8) 

            SAY PGM OWNER CONTOKEN TIMESTMP DSN 

            rec_found = rec_found + 1 

         end 

         when SQLCODE = 100 then leave /* Nichts mehr gefunden */ 

         otherwise do 

            call sqlCA 

            call abend "Abbruch wegen SQL-Fehlers." 

         End 

      End 

   End 

   "EXECSQL CLOSE C1" 

Return 

 

SQLCA: 

   SAY 'SQLSTMT' SQLSTMT 

   SAY 'SQLSTATE=' SQLSTATE 

   SAY 'SQLWARN =' 

   Do §=0 to 10 

      SAY SQLWARN.§ 

   End 

 

   SAY 'SQLERRD =' 

   Do §=0 to 6 

      SAY SQLERRD.§ 

   End 

 

   SAY 'SQLERRP  =' SQLERRP 

   SAY 'SQLERRMC =' SQLERRMC 

   SAY 'SQLCODE  =' SQLCODE 

   SAY 'SQLSTMT  =' SQLSTMT 

RETURN 0 

 

abend: procedure 

   arg zedlmsg 

   address ispexec "setmsg msg(isrz000)" 

exit 12 

6.2.3. Eine kleine REXX-ISPF-DB2-Anwendung 

Hier wollen wir uns anschauen, was im Alltagsleben häufig vorkommt. 

▪ In einer bestehenden Datenbank soll ein Tablespace und eine Tabelle angelegt werden. Für 
den Fall, dass dies wiederholt werden soll (Feld vergessen oder falsch definiert), soll dies 
wiederholbar sein 

▪ Initiales Befüllen der Tabelle aus Datei 
▪ ISPF-Dialog, der die Daten aus DB2 liest und über eine temporäre ISPF-Tabelle anzeigt. 

Dort soll möglich sein, Sätze einzufügen, zu ändern, anzuzeigen und zu löschen. Verände-
rungen sollen im Hintergrund im DB2 nachgezogen werden. 

6.2.3.1. Erstellen eines Tablespace mit Tabelle 

Wir unterstellen, dass die Datenbank bereits existiert. 

/* REXX 

 

   Programmbeschreibung und Update History am Ende 

*/ 

parse source . RxTyp RxName . 

Signal on NoValue Name Event_Handler 

Signal on Syntax  Name Event_Handler 

Main_Start: 

   Call Set_Vars 

   Call Activate_DSNRexx 

   address dsnrexx 

   Call Connect_to_DB2 

   Call Set_Curr_SQLID 

   Call Drop_Table_Space 

   Call Create_Table_Space 

   Call Create_Table 

   Call Load_Table 

   Call DisConnect_from_DB2 

   Call DeActivate_DSNRexx 

Main_Ende: 

 

exit 0 

 

Set_Vars: 

   /****************************************************************** 

   * Variablendefinitionen                                           * 

   ******************************************************************/ 

   Foreground = sysvar("SYSENV") = "FORE" 

   Batch      = ^Foreground 

   Db2Sys     = "DB2Z" 

   Db2Lib     = "ABC.DB20000.APFLOAD" 

   DataBase   = "DAZEDU00" 

   TableSpace = "SAZEDU00" 

   Schema     = UserID() 

   Table      = UserID()"A" 

   SQLSTMT    =  

Return 

 

Set_Curr_SQLID: 

   /****************************************************************** 

   * SQLID Setzen                                                    * 

   ******************************************************************/ 

   SQLSTMT = "SET CURRENT SQLID = 'GRP4711'" 

   "EXECSQL" SQLSTMT 

   if SQLCode <> 0 then call SQLCA 

return 

 

Drop_Table_Space: 

   /****************************************************************** 

   * Tablespace löschen                                              * 

   ******************************************************************/ 

   SQLSTMT = "DROP TABLESPACE" DataBase"."TableSpace 

   "EXECSQL" SQLSTMT 

   if SQLCode <> 0 then call SQLCA 

   "EXECSQL COMMIT" 

   if SQLCode <> 0 then call SQLCA 

return 

 

Create_Table_Space: 

   /****************************************************************** 

   * Tablespace erstellen                                            * 

   ******************************************************************/ 

   SQLSTMT = "" , 

      'CREATE TABLESPACE' TableSpace 'IN' DataBase , 

      'USING STOGROUP "DATA"' , 

      'PRIQTY  -1      ' ,  /* Primary Space für DB2-Dataset        */ 

      'SECQTY  -1      ' ,  /* Secondary Space für DB2-Dataset      */ 

      'PCTFREE 15      ' ,  /* Freier Speicher in % nach reorg      */ 

      'MAXPARTITIONS 1 ' ,  /* Max Anzahl Partitionen Tablespace    */ 

      'CCSID EBCDIC    ' ,  /* Encodung Type (Default ist ASCII)    */ 

      '' 

   "EXECSQL" SQLSTMT 

   if SQLCode < 0 then , 

      call SQL_Err "CREATE TABLESPACE ist fehlgeschlagen." 

   "EXECSQL COMMIT" 

   if SQLCode < 0 then call SQL_Err "COMMIT ist fehlgeschlagen." 

Return 

 

Create_Table: 

   /****************************************************************** 

   * Tabelle erstellen                                               * 

   ******************************************************************/ 

   SQLSTMT = "CREATE TABLE" Schema"."Table "(" , 

               "Vorname       CHAR(25)," , 

               "Nachname      CHAR(30)," , 

               "PLZ           CHAR(05)," , 

               "Ort           CHAR(30)," , 

               "Strasse       CHAR(30)," , 

               "Geburts_Datum CHAR(10)," , 

               "Pseudonym     CHAR(20)," , 

               "Hobby         CHAR(30)," , 

               "Abteilung     CHAR(20)," , 

               "TIMESTMP      TIMESTAMP NOT NULL" , 

            ")" , 

         "IN" DataBase"."TableSpace , 

         "CCSID EBCDIC" 

   "EXECSQL "SQLSTMT 

   if SQLCode < 0 then call SQL_Err "CREATE TABLE ist fehlgeschlagen." 

   "EXECSQL COMMIT" 

   if SQLCode < 0 then call SQL_Err "COMMIT ist fehlgeschlagen." 

Return 

 

Load_Table: 

   /****************************************************************** 

   * Befüllen der DB2-Tabelle                                        * 

   * Einlesen des Memberinhalts in den Stem AdrDb2.                  * 

   * Parsen der Zeileninhalte.                                       * 

   * Insert in die DB2-Tabelle.                                      * 

   ******************************************************************/ 

   address tso 

   "alloc dd(AdrDb2) da('SP1.EDU.DATA(ADRDB2)') shr reuse" 

   "execio * diskr AdrDb2 (stem AdrDb2. finis" 

   "Free dd(AdrDb2)" 

   DB2Flds    = "(" !!           , 

                "Vorname,"       ,        /*  Vorname               */ 

                "Nachname,"      ,        /*  Nachname              */ 

                "PLZ,"           ,        /*  Postleitzahl          */ 

                "Ort,"           ,        /*  Ortname               */ 

                "Strasse,"       ,        /*  Straßee               */ 

                "Geburts_Datum," ,        /*  Geb_Datum             */ 

                "Pseudonym,"     ,        /*  Pseudonym             */ 

                "Hobby,"         ,        /*  Hobby                 */ 

                "Abteilung,"     ,        /*  Abteilung             */ 

                "TIMESTMP"       ,        /*  TIMESTMP              */ 

                !! ")" 

 

   address DSNRexx 

   do §=1 to AdrDb2.0 

      parse var AdrDb2.§ Vn  "!" Nn     "!" Plz    "!" Ort    "!" , 

                         Str "!" GebDat "!" Pseudo "!" Hobby  "!" Abt 

      DB2Vals  = "(" !! , 

                 "'"Vn"',"          ,     /*  Vorname               */ 

                 "'"Nn"',"          ,     /*  Nachname              */ 

                 "'"Plz"',"         ,     /*  PLZname               */ 

                 "'"Ort"',"         ,     /*  Ortname               */ 

                 "'"Str"',"         ,     /*  Straßee               */ 

                 "'"GebDat"',"      ,     /*  Geb_Datum             */ 

                 "'"Pseudo"',"      ,     /*  Pseudonym             */ 

                 "'"Hobby"',"       ,     /*  Hobbyme               */ 

                 "'"Abt"',"         ,     /*  Abteilung             */ 

                 "'"Time_Stamp()"'" ,     /*  TIMESTMP              */ 

                 ")" 

 

      /*************************************************************** 

      * Einfügen der Daten in die Tabelle                            * 

      ***************************************************************/ 

      SQLSTMT = "INSERT INTO" Schema"."Table DB2Flds "VALUES" DB2Vals 

      "EXECSQL" SQLSTMT 

      if SQLCode <> 0 then , 

         Call SQL_Err "INSERT in" Schema"."Table "ist fehlgeschlagen." 

   End 

   /****************************************************************** 

   * Alle Änderungnen in der Datenbank dauerhaft machen.             * 

   ******************************************************************/ 

   "EXECSQL COMMIT" 

   if SQLCode <> 0 then Call SQL_Err "Commit ist fehlgeschlagen." 

Return 

 

Time_Stamp: 

return translate("CcJj-Mm-Dd",date("S"),"CcJjMmDd") !! "-" !! , 

       translate(time("L"),".",":") 

 

Activate_DSNRexx: 

   /****************************************************************** 

   * Allocieren der DB2Load-Lib                                      * 

   * Starten des DSNRexx Befehlsenvironments.                        * 

   ******************************************************************/ 

   call LIBADD "SYSDB.DB20000.DSNLOAD" 

   DBRc = RxSubCom('ADD','DSNREXX','DSNREXX') 

   if DBRc <> 0 then Call Abend "DSNRexx konnte nicht geladen werden!" 

return 

 

DeActivate_DSNRexx: 

   /****************************************************************** 

   * Abbauen des DSNRexx Befehlsenvironments.                        * 

   * DeAllocieren der DB2Load-Lib                                    * 

   ******************************************************************/ 

   DBRC = RxSubCom('DELETE','DSNREXX','DSNREXX') 

   call LIBRESET "SYSDB.DB20000.DSNLOAD" 

return 

 

Connect_to_DB2: 

   /****************************************************************** 

   * Verbindung zum DB2 Aufbauen.                                    * 

   ******************************************************************/ 

   "CONNECT" Db2Sys 

   if SQLCode <> 0 then , 

      Call SQL_Err "CONNECT zu" Db2Sys "ist fehlgeschlagen." 

Return 

 

DisConnect_from_DB2: 

   /****************************************************************** 

   * Connect zum DB2 Abbauen.                                        * 

   ******************************************************************/ 

   "DISCONNECT" 

   if SQLCode <> 0 then call sqlca 

return 

 

SQL_Err: 

   /****************************************************************** 

   * Ausgabe der SQLCA, Disconnect, Abbau DSNREXX-Environment        * 

   ******************************************************************/ 

   parse arg ErrMsg 

   say ErrMsg 

   Call SQLCA 

   Call DisConnect_from_DB2 

   Call DeActivate_DSNRexx 

   Call Abend ErrMsg 

Return 

 

abend: 

   /****************************************************************** 

   * Ausgabe der ZEDLMSG im Fehlerfall                               * 

   ******************************************************************/ 

   parse arg zedlmsg 

   parse source . . RxN . 

   if Batch then say RxN"-E" zedlmsg 

   else          address ispexec "setmsg msg(isrz000)" 

exit 99 

 

SQLCA: 

   /****************************************************************** 

   * Ausgabe der Informarionen zum SQL-Returncode                    * 

   ******************************************************************/ 

   say "SQLSTMT  ="  SQLSTMT 

   say "SQLCODE  = >"SQLCODE"<" 

   say "SQLSTATE = >"SQLSTATE"<" 

   say "SQLERRMC = >"SQLERRMC"<" 

   say "SQLERRP  = >"SQLERRP"<" 

 

   do _1 = 1 to 10 

      say left("SQLWARN."_1"= >",13) !! SQLWARN._1 !! "<" 

   end 

   do _1 = 1 to 6 

      say left("SQLERRD."_1"= >",13) !! SQLERRD._1 !! "<" 

   end 

return 

 

Event_Handler: 

   /****************************************************************** 

   * Behandlung diverser Events                                      * 

   ******************************************************************/ 

   parse source . . RxN . 

   CondC = Condition("C") 

   CondD = Condition("D") 

   CondS = Condition("S") 

   CondI = Condition("I") 

   Select 

      when CondC = "NOVALUE" then do  

         say RxN"-E" "NOVALUE Fehler in Zeile" , 

             SigL", undefinierte Variable: "CondD 

      End 

      when CondC = "SYNTAX" then do 

         say RxN"-E" "SYNTAX Fehler in Zeile "SigL 

         say RxN"-E" "RC = "rc", "errortext(rc) 

         say RxN"-E" "Desc: "CondD 

      end 

      when wordpos(CondC,"ERROR FAILURE") > 0 then do 

         say RxN"-E" CondC": RC = "rc" in Zeile "SigL 

         say RxN"-E" "Statement: "CondD 

      end 

      otherwise do 

         say RxN"-E" CondC"-Fehler in Zeile "SigL 

         say RxN"-E" "COND(D): "CondD 

         say RxN"-E" "COND(S): "CondS 

         say RxN"-E" "COND(I): "CondI 

         say RxN"-E" "Return Code = "rc 

      end 

   end 

   if sourceline() > 0 then, 

      say right(SigL,6)": "left(sourceline(SigL),72) 

   say RxN"-E" "Rexx endet mit RC = 16" 

   if SYSVAR("SYSISPF") = "ACTIVE" then do 

      zispfrc = 16 

      address ispexec "VPUT (ZISPFRC) SHARED" 

   end 

exit 16 

/* Programmbeschreibung ********************************************** 

 

   Nutzung als..: Command 

   Zweck........: Erstellen der Tabelle für DB2APPL 

   Parameter....: Keine 

   Rückgaben....:  0 alles ist OK 

                  16 Syntaxfehler oder Novalue 

                  99 Abend 

   Beschreibung.: Verbindung zu DB2 herstellen 

                  Tablespace aus der angegebenen Database entfernen 

                  Tablespace in der angegebenen Database erstellen 

                  Tabelle erstellen 

                  Tabelle befüllen 

                  Verbindung zu DB2 trennen 

 

-- Update History----------------------------------------------------- 

 

   09.02.2021 Karlheinz Wittemann Erstellung 

 

   tt.mm.jjjj Name 

              ° Grund 

              ° Grund 

 

*/ 

6.2.3.2. Das Dialogprogramm 

Steuert den ISPF-Dialog. Zu Beginn werden die Daten aus dem DB2 gelesen und in einer temporä-
ren ISPF-Tabelle abgelegt. Das Programm zeigt diese Daten an und erlaubt dem Anwender die 
Gängigen Funktionen für die Datensätze (anzeigen, ändern, löschen, einfügen). 

Sofern die Änderungen im DB2 geklappt haben, werden sie auch in der temporären ISPF-Tabelle 
nachgezogen. 

/* REXX * Programmbeschreibung und Update History am Ende 

*/ 

parse source . RxTyp RxName . RxLib . 

Panlib = left(RxLib,lastpos(".",RxLib))"PANELS" 

 

Signal on NoValue Name Event_Handler 

Signal on Syntax  Name Event_Handler 

 

Main_Start: 

   Address Ispexec 

   "control errors return" 

   "libdef ispplib dataset id('&Panlib') stkadd" 

   call Set_Vars 

   call Activate_DSNRexx 

   call Connect_to_DB2 

   call Set_Curr_SQLID 

   call Create_ISPF_Table 

   call Fill_ISPF_Table 

   call Show_ISPF_Table 

   call Free_ISPF_Table 

   call DeActivate_DSNRexx 

   "libdef ispplib" 

Main_Ende: 

Exit 

 

Set_Vars: 

   /****************************************************************** 

   * Variablendefinitionen                                           * 

   ******************************************************************/ 

   read_from_ds   = abbrev("DATEI",translate(arg(1)),1) ! , 

                    abbrev("DSN"  ,translate(arg(1)),1) 

   DsnRxTab       = userid()"A" 

   Db2Sys         = "DB2Z" 

   Db2Lib         = "SYSDB.DB20000.APFLOAD" 

   Schema         = UserID() 

   Table          = UserID()"A" 

   SQLSTMT        =  

 

   Fields         = "" , 

                    "Vorname       :Vn       " , 

                    "Nachname      :Nn       " , 

                    "PLZ           :Plz      " , 

                    "Ort           :Ort      " , 

                    "Strasse       :Str      " , 

                    "Geburts_Datum :GebDat   " , 

                    "Pseudonym     :Pseudo   " , 

                    "Hobby         :Hobby    " , 

                    "Abteilung     :Abt      " , 

                    "TIMESTMP      :TimeStmp " , 

                    "" 

   ISPFVars       = 

   Db2Flds        = 

   do while fields <> "" 

      parse var fields DB2Feld ISPFVar fields 

      ISPFVars = ISPFVars ISPFVar 

      Db2Flds  = Db2Flds  DB2Feld 

   End 

   ISPFVars = space(ISPFVars,1,",") 

   Db2Flds  = space(Db2Flds ,1,",") 

 

   LCmds          = "I,U,B,D" 

   TabKeys        = "" 

   TabNames       = "" !! ,      /* Feldnamen der ISPF-Tabelle      */ 

                    "Vn"       , /* Vorname                         */ 

                    "Nn"       , /* Nachname                        */ 

                    "Plz"      , /* Postleitzahl                    */ 

                    "Ort"      , /* Ort                             */ 

                    "Str"      , /* Straße                          */ 

                    "GebDat"   , /* Geburtsdatum                    */ 

                    "Pseudo"   , /* Pseudonym                       */ 

                    "Hobby"    , /* Hobby                           */ 

                    "Abt"      , /* Abteilung                       */ 

                               , /* Technische Felder               */ 

                    "TimeStmp" , /* TimeStamp X(26)                 */ 

                               , /* 2021-03-09-08.02.00.000000      */ 

                    "VnUc"     , /* Vorname    Upper Case           */ 

                    "NnUc"     , /* Nachname   Upper Case           */ 

                    "OrtUc"    , /* Ort        Upper Case           */ 

                    "StrUc"    , /* Straße     Upper Case           */ 

                    "GebDatG"  , /* Geburtsdatum German tt.mm.jjjj  */ 

                    "PseudoUc" , /* Pseudonym  Upper Case           */ 

                    "HobbyUc"  , /* Hobby      Upper Case           */ 

                    "AbtUc"    , /* Abteilöung Upper Case           */ 

                    !! "" 

 

   Sort.          = "NnUc,C,A,VnUc,C,A"                 /* Nachname */ 

   Sort.V         = "VnUc,C,A,NnUc,C,A"                 /* Vorname  */ 

   Sort.H         = "HobbyUc,C,A,NnUc,C,A"              /* Hobby    */ 

   Sort.P         = "PseudoUc,C,A,NnUc,C,A"             /* Pseudonym*/ 

   Sort.A         = "AbtUc,C,A,NnUc,C,A,VnUc,C,A"       /* Abteilung*/ 

 

   Header.        = "Unbekannte Funktion"               /* Unbekannt*/ 

   Header.I       = "Einfügen Datensatz"                /* Insert   */ 

   Header.D       = "Löschen Datensatz"                 /* Delete   */ 

   Header.U       = "Ändern Datensatz"                  /* Update   */ 

   Header.B       = "Anzeigen Datensatz"                /* Browse   */ 

Return 

 

Set_Curr_SQLID: 

   /****************************************************************** 

   * Die SQL-Aufrufe laufen mit den Rechten von GRP4711              * 

   ******************************************************************/ 

   SQLSTMT = "SET CURRENT SQLID = 'GRP4711'" 

   "EXECSQL" SQLSTMT 

   if SQLCode <> 0 then call SQLCA 

return 

 

Create_ISPF_Table: 

   "tbend &DsnRxTab" 

   "tbcreate &DsnRxTab nowrite keys(&TabKeys) names(&TabNames)" 

   if RC > 0 then call abend "RC" RC "bei TbCreate DsnRxTab" 

return 

 

Free_ISPF_Table: 

   "TbEnd &DsnRxTab"  

Return 

 

Fill_ISPF_Table: 

   if read_from_ds then call Fill_from_Dataset 

   else                 call Fill_from_DB2 

return 

 

Fill_from_DB2: 

   /****************************************************************** 

   * Befüllen der ISPF-Tabelle aus dem DB2                           * 

   * Definieren des Cursors für die Ergebnistabelle der Select-      * 

   * Anweisung.                                                      * 

   * FETCH verschiebt den Cursor und stellt den Inhalt der aktuelen  * 

   * Zeile der Ergebnistabelle in die Rexx-Variablen.                * 

   ******************************************************************/ 

   address DSNrexx 

   SQLSTMT = "SELECT" DB2Flds "FROM" Schema"."Table";" 

   "EXECSQL PREPARE S1 FROM :SQLSTMT" 

   "EXECSQL DECLARE C1 CURSOR FOR S1" 

   "EXECSQL OPEN C1" 

   do § = 1 

      "EXECSQL FETCH C1 INTO" ISPFVars 

      if SQLCode = 100 then leave 

      if SQLCode <> 0  then Call SQL_Err "FETCH ist fehlgeschlagen." 

      call Convert_Fields 

      address ispexec "TbAdd &DsnRxTab order" 

   end 

   "EXECSQL CLOSE C1" 

Return 

 

Fill_from_Dataset: 

   address tso 

   "alloc dd(AdrDb2) da('SP1.EDU.DATA(ADRDB2)') shr reuse" 

   "execio * diskr AdrDb2 (stem AdrDb2. finis" 

   address ispexec 

   do §=1 to AdrDb2.0 

      parse var AdrDb2.§ Vn  "!" Nn     "!" Plz    "!" Ort    "!" , 

                         Str "!" GebDat "!" Pseudo "!" Hobby  "!" Abt 

      TimeStmp = Time_Stamp() 

      call Convert_Fields 

      "tbadd &DsnRxTab" 

   End 

Return 

 

Convert_Fields: 

   VnUc     = translate(Vn) 

   NnUc     = translate(Nn) 

   OrtUc    = translate(Ort) 

   StrUc    = translate(Str) 

   GebDatG  = translate("Tt.Mm.CcJj",gebdat,"CcJjMmTt") 

   PseudoUc = translate(Pseudo) 

   HobbyUc  = translate(Hobby) 

   AbtUc    = translate(Abt) 

Return 

 

Time_Stamp: 

return translate("CcJj-Mm-Dd",date("S"),"CcJjMmDd") !! "-" !! , 

       translate(time("L"),".",":") 

/* 2004-03-09-08.02.00.000000   >> Timestamp-Format */                         

 

Show_ISPF_Table: 

   call PCmd_Sort "N" 

   CRP = 1 

   do forever 

      "TbTop   &DsnRxTab" 

      "TbSkip  &DsnRxTab number(&CRP)" 

      "TbDispl &DsnRxTab panel(Db2PanT)" 

      CRP = ZTdTop 

 

      if rc > 4 then leave 

      parse upper var ZCmd PCmd Parms 

      if PCmd <> "" then do 

         select 

            When abbrev("INSERT",PCmd,1) then call LCmd_BDIU "I" 

            When abbrev("RESET" ,PCmd,3) then Show = "ALL" 

            When abbrev("SORT"  ,PCmd,1) then call PCmd_Sort parms 

            When abbrev("SCAN"  ,PCmd,2) then call PCmd_Scan parms 

            Otherwise Nop 

         End 

      End 

      else do 

         do while zTdSels > 0 

            "Control Display Save" 

            select /* nicht üblich aber effizient ;-) */ 

               when wordpos(LCmd,"B D I U") > 0 then call LCmd_BDIU lcmd 

               otherwise Nop 

            end 

            "Control Display Restore" 

            if ZTdSels > 1 then "TbDispl &DsnRxTab" 

            else                ZTdSels = 0 

         end 

      end 

   end 

return 

 

LCmd_BDIU: 

   arg Fkt 

   if Show_Panel(Fkt) > 0 then return 

   select 

      when Fkt = "I" then Call LCmd_Insert 

      when Fkt = "U" then Call LCmd_Update 

      when Fkt = "D" then Call LCmd_Delete 

      otherwise nop 

   end 

return 

 

Show_Panel: 

   if Fkt = "I" then "TbvClear &DsnRxTab" 

   zwinttl = header.Fkt 

   drop zcmd 

   "addpop row(1) column(10)" 

   "display panel(DB2PAND)" 

   DispRC = RC 

   "rempop" 

   call Convert_Fields 

return DispRC 

 

LCmd_Insert: 

   /****************************************************************** 

   * Einfügen eines neuen Eintrags.                                  * 

   ******************************************************************/ 

   TimeStmp = Time_Stamp() 

   DB2Vals  = "" !! , 

      "'"strip(Vn      )"'," ,   /*  Vorname                 */ 

      "'"strip(Nn      )"'," ,   /*  Nachname                */ 

      "'"strip(Plz     )"'," ,   /*  PLZname                 */ 

      "'"strip(Ort     )"'," ,   /*  Ortname                 */ 

      "'"strip(Str     )"'," ,   /*  Straßee                 */ 

      "'"strip(GebDat  )"'," ,   /*  Geb_Datum               */ 

      "'"strip(Pseudo  )"'," ,   /*  Pseudonym               */ 

      "'"strip(Hobby   )"'," ,   /*  Hobbyme                 */ 

      "'"strip(Abt     )"'," ,   /*  Abteilung               */ 

      "'"strip(TimeStmp)"'" ,    /*  TIMESTMP                */ 

      "" 

 

   SQLSTMT = "INSERT INTO" Schema"."Table "("DB2Flds")" , 

                 "VALUES ("DB2Vals")" 

   address dsnrexx "EXECSQL" SQLSTMT 

   if SQLCode <> 0 then Call SQLCA 

   else do 

      address dsnrexx "EXECSQL COMMIT" 

      "tbadd &DsnRxTab order" 

   End 

Return 

 

LCmd_Update: 

   /****************************************************************** 

   * Updaten eines bestehenden Eintrags.                             * 

   ******************************************************************/ 

   NewTStmp = Time_Stamp() 

   SQLSTMT = "UPDATE" Schema"."Table , 

      "SET VORNAME"       !! "='" !! Vn       !! "'," , 

          "NACHNAME"      !! "='" !! Nn       !! "'," , 

          "PLZ"           !! "='" !! Plz      !! "'," , 

          "ORT"           !! "='" !! Ort      !! "'," , 

          "STRASSE"       !! "='" !! Str      !! "'," , 

          "GEBURTS_DATUM" !! "='" !! GebDat   !! "'," , 

          "PSEUDONYM"     !! "='" !! Pseudo   !! "'," , 

          "HOBBY"         !! "='" !! Hobby    !! "'," , 

          "ABTEILUNG"     !! "='" !! Abt      !! "'," , 

          "TIMESTMP"      !! "='" !! NewTStmp !! "'"  , 

      "WHERE TIMESTMP='"TimeStmp"'" 

      /* 

      Es würde auch ein WHERE TIMESTAMP=:TimeStamp gehen. 

      (und natürlich auch für die anderen Felder in der SET-Anweisung) 

      */ 

   address dsnrexx "EXECSQL" SQLSTMT 

   select 

      when SQLCode = 100 then do 

         zedlmsg = "Satz wurde inzwischen gelöscht oder geändert" , 

                      "(Satz wird aus Tabelle entfernt)." 

         "setmsg msg(isrz000)" 

         "tbdelete &DsnRxTab" 

      end                                                                      

      when SQLCode <> 0  then Call SQLCA 



Inside z/OS Seite 370 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

      otherwise do 

         address dsnrexx "EXECSQL COMMIT" 

         TimeStmp = NewTStmp 

         "tbput &DsnRxTab" 

      End 

   End 

Return 

 

LCmd_Delete: 

   /****************************************************************** 

   * Löschen eines Eintrags.                                         * 

   ******************************************************************/ 

   SQLSTMT = "DELETE" Schema"."Table "WHERE TIMESTMP = '"TimeStmp"'" 

   address dsnrexx "EXECSQL" SQLSTMT 

   address dsnrexx "EXECSQL COMMIT" 

   "tbdelete &DsnRxTab" 

Return 

 

PCmd_Sort: 

   parse value arg(1) with sfld 2 . 

   SFlds = Sort.Sfld 

   "tbsort &DsnRxTab fields(&SFlds)" 

Return 

 

PCmd_Scan: 

   arg scanparms 

   if scanparms = "" then do 

      Show = "ALL" 

      Return 

   End 

   if words(scanparms) // 2 > 0 then do 

      zedlmsg = "Für Scan immer 2er Päärchen angeben" , 

                "(col wert [col wert]...)" 

      "setmsg msg(isrz000)" 

      Return 

   End 

   "tbvclear &DsnRxTab" 

   do while scanparms <> "" 

      parse var scanparms col val scanparms 

      select 

         when left(col,1) = "V" then VnUc     = val 

         when left(col,1) = "N" then NnUc     = val 

         when left(col,1) = "G" then GebDat   = val 

         when left(col,1) = "P" then PseudoUc = val 

         when left(col,1) = "H" then HobbyUc  = val 

         when left(col,1) = "A" then AbtUc    = val 

         otherwise nop 

      end 

   end 

   "tbsarg &DsnRxTab" 

   Show = "SCAN" 

Return 

 

Activate_DSNRexx: 

   /****************************************************************** 

   * Allocieren der DB2Load-Lib                                      * 

   * Starten des DSNRexx Befehlsenvironments.                        * 

   ******************************************************************/ 

   Call LIBADD Db2Lib 

   DBRc = RxSubCom('ADD','DSNREXX','DSNREXX') 

   if DBRc <> 0 then Call Abend "DSNRexx konnte nicht geladen werden!" 

return 

 

DeActivate_DSNRexx: 

   /****************************************************************** 

   * Abbauen des DSNRexx Befehlsenvironments.                        * 

   * Freigabe der DB2-Loadlib                                        * 

   ******************************************************************/ 

   DBRC = RxSubCom('DELETE','DSNREXX','DSNREXX') 

   Call LIBRESET DB2Lib 

Return 

 

Connect_to_DB2: 

   /****************************************************************** 

   * Connect zum DB2 Aufbauen.                                       * 

   ******************************************************************/ 

   address dsnrexx 

   "CONNECT" Db2Sys 

   if SQLCode <> 0 then , 

      Call SQL_Err "CONNECT zu" Db2Sys "ist fehlgeschlagen." 

Return 

 

DisConnect_from_DB2: 

   /****************************************************************** 

   * Connect zum DB2 Abbauen.                                        * 

   ******************************************************************/ 

   address dsnrexx "DISCONNECT" 

   if SQLCode <> 0 then call sqlca 

return 

 

SQL_Err: 

   /****************************************************************** 

   * Ausgabe der SQLCA, Disconnect, Abbau DSNREXX-Environment        * 

   ******************************************************************/ 

   parse arg ErrMsg 

   say ErrMsg 

   Call SQLCA 

   Call DisConnect_from_DB2 

   Call DeActivate_DSNRexx 

   Call Abend ErrMsg 

Return 

 

abend: procedure expose RxName 

   /****************************************************************** 

   * Ausgabe der ZEDLMSG im Fehlerfall                               * 

   ******************************************************************/ 

   parse arg zedlmsg 

   if sysvar("SYSENV")="FORE" then address ispexec "setmsg msg(isrz000)" 

   else                            say RXN"-E" zedlmsg 

exit 99 

 

SQLCA: 

   /****************************************************************** 

   * Ausgabe SQLCA im Bedarfsfall                                    * 

   ******************************************************************/ 

   Signal off NoValue 

   SAY "SQLSTMT  =" SQLSTMT 

   SAY "SQLSTATE =" SQLSTATE 

   SAY "SQLWARN  =" 

   Do §=0 to 10 

      SAY SQLWARN.§ 

   End 

   SAY "SQLERRD =" 

   Do §=0 to 6 

      SAY SQLERRD.§ 

   End 

   SAY "SQLERRP  =" SQLERRP 

   SAY "SQLERRMC =" SQLERRMC 

   SAY "SQLCODE  =" SQLCODE 

   SAY "SQLSTMT  =" SQLSTMT 

Return 

 

Event_Handler: 

   /****************************************************************** 

   * Behandlung diverser Events                                      * 

   ******************************************************************/ 

   parse source . . RxN . 

   CondC = Condition("C") 

   CondD = Condition("D") 

   CondS = Condition("S") 

   CondI = Condition("I") 

   Select 

      when CondC = "NOVALUE" then do 

         say RxN"-E" "NOVALUE Fehler in Zeile" , 

             SigL", undefinierte Variable: "CondD 

      End 

      when CondC = "SYNTAX" then do 

         say RxN"-E" "SYNTAX Fehler in Zeile "SigL 

         say RxN"-E" "RC = "rc", "errortext(rc) 

         say RxN"-E" "Desc: "CondD 

      end 

      when wordpos(CondC,"ERROR FAILURE") > 0 then do 

         say RxN"-E" CondC": RC = "rc" in Zeile "SigL 

         say RxN"-E" "Statement: "CondD 

      end 

      otherwise do 

         say RxN"-E" CondC"-Fehler in Zeile "SigL 

         say RxN"-E" "COND(D): "CondD 

         say RxN"-E" "COND(S): "CondS 

         say RxN"-E" "COND(I): "CondI 

         say RxN"-E" "Return Code = "rc 

      end 

   end 

   if sourceline() > 0 then, 

      say right(SigL,6)": "left(sourceline(SigL),72) 

   say RxN"-E" "Rexx endet mit RC = 16" 

   if SYSVAR("SYSISPF") = "ACTIVE" then do 

      zispfrc = 16 

      address ispexec "VPUT (ZISPFRC) SHARED" 

   end 

exit 16 

 

/* Programmbeschreibung ********************************************** 

 

   Nutzung als..: Command 

   Zweck........: Applikation zur Veranschaulichung von DSNREXX 

   Parameter....: Keine 

   Rückgaben....: Keine 

 

-- Update History----------------------------------------------------- 

 

   16.05.2021 Karlheinz Wittemann Erstellung 

 

   tt.mm.jjjj Name 

              ° Grund 

              ° Grund 

 

*/ 

6.2.3.3. Das Tabellenpanel 

)ATTR default(%+_) 

   ~ TYPE(INPUT)   caps(ON)       color(RED)    PADC('.') 

   $ TYPE(OUTPUT)  caps(OFF)      color(TURQ) 

   @ TYPE(OUTPUT)  caps(OFF)      color(TURQ) 

   \ TYPE(OUTPUT)  caps(OFF)      color(TURQ) 

   µ TYPE(OUTPUT)  caps(OFF)      color(TURQ) 

   [ TYPE(OUTPUT)  caps(OFF)      color(TURQ) 

   â TYPE(OUTPUT)  caps(OFF)      color(TURQ) 

   ê TYPE(OUTPUT)  caps(OFF)      color(TURQ) 

 

   ² TYPE(TEXT)    hilite(uscore) color(white) 

 

   ] AREA(DYNAMIC) 

   ³ TYPE(CHAR)    color(YELLOW)  hilite(USCORE) 

   } TYPE(CHAR)    color(WHITE)   hilite(USCORE) 

   { TYPE(CHAR)    color(BLUE) 

)BODY expand(§§) 

+-----------%Anzeige der Adress-Tabelle+§-§ 

+Command ==>_ZCMD § §+SCROLL =>_SAMT+ 

+ 

+PrimCMDs: ]PRIMCM,PRIMSH                                            ] 

+LineCMDs: ]LINECM,LINESH                                            ] 

%|+ 

²?²Vorname     ²Nachname        ²Gebdatum  ²Pseudonym²Hobby        ²Abt 

)MODEL clear(LCmd) rows(&SHOW) 

~Z$VN          @NN              \GebDatG   [Pseudo   âHobby        êAbt 

)INIT 

   .zvars = '(LCMD)' 

 

   &PrimCm   = 'Insert  Sort col Scan fld wert' 

   &PrimSh   = '³{{{{{{{³{{{{{{{{³³{{{{{{{{{{{' 

   &LineCm   = 'Insert  Update   Browse   Delete' 

   &LineSh   = '}{{{{{{{}{{{{{{{{}{{{{{{{{}{{{{{' 

   if (&SAMT = &Z)    &SAMT = CSR 

   if (&SHOW ^= SCAN) &SHOW = ALL 

 

   if (&SFld = V) .attrchar($) = 'color(Yellow)' 

   if (&SFld = N) .attrchar(@) = 'color(Yellow)' 

   if (&SFld = H) .attrchar(â) = 'color(Yellow)' 

   if (&SFld = P) .attrchar([) = 'color(Yellow)' 

   if (&SFld = A) .attrchar(ê) = 'color(Yellow)' 

)PROC 

REFRESH(*) 

if (ver  (&LCmd,listv,&LCMDS)) 

else 

   &ZEDLMSG = 'Zeilenkommando ungültig' 

   .MSG = ISRZ000 

)END 

6.2.3.4. Das Datenpanel 

)ATTR default($+_) 

   [ type(input) caps(OFF) color(RED) padc(_) 

)body expand(//) window(55,15) 

+Command=>_ZCMD 

+ 

+Vorname...>[Vn                   +      (Pflichtfeld) 

+Name......>[Nn                      +   (Pflichtfeld) 

+PLZ.......>[Plz  +                      (Pflichtfeld) 

+Ort.......>[Ort                        +(Pflichtfeld) 

+Strasse...>[Str                        +(Pflichtfeld) 

+ 

+Geb.-Datum>[GebDat  + (jjjjmmtt          Pflichtfeld) 

+ 

+Pseudonym.>[Pseudo           +          (Pflichtfeld) 

+Hobby.....>[Hobby            +          (Pflichtfeld) 

+Abteilung.>[Abt         +               (Pflichtfeld) 

)INIT 

   if (&LCmd,EQ,B) .attrchar([)='Type(output) color(blue)' 

   if (&LCmd,EQ,D) .attrchar([)='Type(output) color(Yellow)' 

)REINIT 

)PROC 

if (ver(&Fkt,list,I,U)) 

   ver (&Vn,NB) 

   ver (&Nn,NB) 

   ver (&PLZ,NB) 

   ver (&Ort,NB) 

   ver (&Str,NB) 

   ver (&GebDat,NB,NUM) 

   ver (&Pseudo,NB) 

   ver (&Hobby,NB) 

   ver (&Abt,NB) 

   &jahr  = trunc(&GEBDAT,4) 

   &rest  = .trail 

   &monat = trunc(&Rest,2) 

   &rest  = .trail 

   &tag   = trunc(&Rest,2) 

   ver (&jahr ,range,1901,&zstdyear) 

   ver (&monat,range,01,12) 

   ver (&tag  ,range,01,31) 

&ZWINTTL = &Z 

)END 

7. Job Control Language 

Einen wesentlichen Bestandteil in z/OS stellt die Job Control Language dar. Es ist nicht sinnvoll, 
Alles und Jedes unter TSO zu verarbeiten – nur weil es geht. Das Problem ist die zunehmende 
Intelligenz der Workload Manager. Wird unter TSO eine Verarbeitung gestartet, bei der sehr viele 
Datensätze gelesen/geschrieben (womöglich gar sortiert) werden, wird in aller Regel weder Bild-
schirm noch Tastatur in diesen Vorgang eingebunden sein. Dies ist ein Indiz, dass hier ein Miss-
brauch stattfindet, welcher durch Reduzierung der Systemservices geahndet wird. Im Klartext heißt 
das: Je länger eine Verarbeitung dauert, desto langsamer wird sie. 

Aus diesem Grund ist es empfehlenswert, Verarbeitungen, welche ein großes Datenvolumen zu 
bewältigen haben, nicht unter TSO Foreground, sondern im Batch ablaufen zu lassen. Das geht 
selbstverständlich auch mit REXX-Programmen – bis zu einem gewissen Grad sogar dann, wenn 
sie ISPF-Services aufrufen. 

Wir wollen zunächst einige Randinformationen über das Spoolsystem des z/OS betrachten, ohne 
welches eine Batchverarbeitung nicht möglich wäre. 

7.1. Das Spoolsystem 

Die Verwaltung der Arbeitsaufträge (Jobs) wird nicht vom z/OS selbst, sondern von den Job-Entry-
Subsystemen JES2 oderJES3 übernommen. Aus diesem Grund ist ein OS/390 ohne Spoolsystem 
entgegen vielen anderen Betriebssystemen nicht arbeitsfähig. 

Ein Arbeitsauftrag an die DV-Anlage wird durch Steueranweisungen in einer eigens dafür bestimm-
ten Sprache (JCL = Job-Control-Language) beschrieben. 

Um einen reibungslosen Verarbeitungsablauf zu gewährleisten, trägt das Spoolsystem im Einzelnen 
bei: 

▪ Einlesen der Arbeitsaufträge (Jobs) 

▪ Verwalten der Auftragswarteschlangen 

▪ Vorbereitende Arbeiten und Übersetzen von Job-Control-Anweisungen 

▪ Bereitstellen von Prozeduren aus der SYS1.PROCLIB 

▪ Verwalten von Spool-Datenbeständen 

▪ Durchführung der Systemausgaben. 

7.1.1. Internal Reader 

Früher wurden Jobs in Form von 80-stelligen Lochkarten über einen Kartenleser (Reader) eingele-
sen. Als Ausgabemedium standen dem Spoolsystem Stanzer (Puncher) und Drucker (Printer) zur 
Verfügung. 

Heute werden mit Hilfe des TSO-Betriebs über den ISPF-Editor 80-stellige JCL-Anweisungen über 
den INTERNAL-READER (Programm, das den Kartenleser simuliert) eingelesen. INTERNAL-REA-
DER können auch für Kassetten- oder Platteneinheiten gestartet werden, falls Jobs von Kassetten 
oder Platten gelesen werden sollen. 

7.1.2. External Writer 

Die Durchführung der Systemausgabe (SYSOUT) erfolgt standardgemäß über Drucker oder Stan-
zer. Soll die Ausgabe auf ein anderes Ausgabemedium (Magnetplatte, Band, oder ähnliches) erfol-
gen, geschieht dies über den External Writer. 

7.1.3. Prioritäten 

JES verwaltet Jobs in einer Warteschlange (Job Queue). Alle eingelesenen Jobs werden dabei in 
einer Spooldatei zwischengespeichert (Spooling). Wie die Jobs der Reihe nach abgearbeitet wer-
den, kann durch Prioritätsangaben in der Job-Anweisung  (PRTY=zahl)  geregelt werden. Die maxi-
male Prioritätsangabe ist 13, die niedrigste ist 0. Bei gleichen Prioritätsangaben entscheidet JES im 
Allgemeinen nach dem Motto FIFO (First In First Out), oder "Wer zuerst kommt, malt zuerst"  die 
Reihenfolge der Auswahl. 

7.1.4. Job Auswahl 

Die Einleitung zur Ausführung von Jobs geschieht durch den Initiator. Pro Initiator kann immer nur 
ein einziger Job laufen. Theoretisch können 1000 Initiator definiert werden (Default = 3 Initiator). Der 
Initiator fragt ständig nach, ob Jobs zur Ausführung anstehen. Er wählt zu seinen Job-Klassen die 
zugehörigen Jobs aus, ordnet die notwendigen Datenbestände, Datenträger und E/A-Einheiten zu 
und führt das aufgerufene Programm aus. 

7.1.5. JES2 oder JES3? 

Diese Frage stellt sich heute nicht mehr. IBM stellt das JES3 ein. Schrittweise werden immer mehr 
"unverzichtbare" Funktionen ehemaliger JES3-Kunden auch im JES2 angeboten. JES3 wird nicht 
mehr weiterentwickelt. 

7.1.6. Schaubild Durchlaufphasen JES2 

 

7.1.7. JES-Durchlaufphasen 

JES2 durchläuft die Phasen 

▪ Input 

▪ Conversion 

▪ Job Scheduling for Execution 

▪ Output 

▪ Purge 

7.1.7.1. Input-Phase 

Hier erfolgt zunächst das Einlesen der Jobs. Im Allgemeinen wird der Vorgang des Lochkartenlesens 
durch den INTERNAL READER simuliert. Eine eindeutige Jobidentifikationsnummer wird zugeord-
net. Der Job wird in den SPOOL-IN-Bereichen zwischengespeichert, wobei eine Trennung zwischen 
reiner JCL und Instream-Datenbeständen erfolgt.  Die JCL-Sätze werden zur Formalüberprüfung an 
die 

7.1.7.2. Conversion-Phase 

übergeben. Anschließend wird die im Job über Prozeduren aufgerufene JCL eingearbeitet. Alle JCL-
Anweisungen werden in Kontrollblöcke (SWA-Blocks) übersetzt und wieder in der Spooldatei abge-
legt. 

7.1.7.3. Interpretation-Phase 

Diese Phase existiert nur bei JES3 und stellt eine Erweiterung der Conversion-Phase dar. Im JES2 
wird diese Phase erst in der Execution-Phase durch den Initiator wahrgenommen. Bereits hier wird 
festgestellt, welche Datenbestände, Datenträger und E/A-Einheiten benötigt werden. Es werden die 
Steuerblöcke für den Job-Scheduler gebildet (SWA-Kontrollblöcke) und die Zwischenspeicherung 
dieser Blöcke in die Spooldatei vorgenommen.  

7.1.7.4. Execution Phase 

Die Execution-Phase besteht aus mehreren Aktivitäten des Initiators (eigener Adressraum pro Initi-
ator). Der Initiator fordert vom JES über die CSA einen Job an. Entsprechend der Jobklasse und der 
Auswahlpriorität wird der Job aus der Job-Queue ausgewählt. 

7.1.7.5. Interpreter 

Diese Aufgabe wird an dieser Stelle nur im JES2 benötigt (Conversion/Interpretation-Phase bei 
JES3). Es werden Kontrollblöcke für die Dauer eines Jobs in der SWA angelegt. 

7.1.7.6. Allocator 

Auch diese Funktion betrifft an dieser Stelle nur JES2 (Main Device Scheduling bei JES3). Die Da-
teizuweisung wird durchgeführt. Dabei wird geprüft, ob eine Datei bereits von einem anderen Job 
angesprochen wurde. Je nachdem was in der JCL-Anweisung (DISP-Parameter) angegeben wurde, 
wird eine Datei nur von einem Job exklusiv benutzt oder kann von mehreren Jobs parallel angespro-
chen werden. Bei exklusiver Angabe bekommt der Operator die Meldung "Job Waiting for Datasets". 

Zusätzlich wird beim Allocator die Gerätezuweisung der benötigten E/A-Einheiten getroffen. Bei 
JES2 findet die Zuordnung immer stepweise statt. Sind benötigte Datenträger nicht montiert, wird 
der Operator aufgefordert dies zu tun. Bleibt der Job in dieser Phase hängen, weil momentan nicht 
genügend E/A-Geräte zur Verfügung stehen, erhält der Operator über entsprechende Kommandos 
die Möglichkeit, den Job in Wartezustand zu versetzen, oder abzubrechen. 

7.1.7.7. Program Fetch/Program Execution 

Die Programme der einzelnen Jobsteps werden geladen und ausgeführt. 

7.1.7.8. Deallocating Datasets und Devices for Jobsteps. 

Ist ein Step beendet, werden alle Datenbestände, Datenträger und E/A-Einheiten, die in darauffol-
genden Steps nicht mehr gebraucht werden, freigegeben. Bei JES2 findet eine erneute Zuordnung 
von Dateien, Datenträgern und E/A-Einheiten bei jedem weiteren Step statt. Bei JES3 wird der 
nächste Step ausgeführt. 

7.1.7.9. Deallocating Datasets und Devices for Job 

Nach Beendigung des Jobs (alle Steps) werden alle Dateien, Datenträger und E/A-Einheiten freige-
geben. 

7.1.7.10. Job Termination Request 

Das Ende eines Jobs wird JES mitgeteilt. Der Initiator übergibt die Steuerung über den Job an das 
JES zurück. 

7.1.7.11. Output-Phase 

Die Ausgabe erfolgt normalerweise auf einen Drucker. Soll die Ausgabe über Bänder, Kassetten 
oder Plattenspeicher erfolgen, muss dies über den EXTERNAL WRITER erfolgen. 

Die Ausgabe kann aber auch an lokale oder andere über Leitung verbundene E/A-Geräte ausgege-
ben werden (RJE = Remote Job Entry) oder sogar an andere Rechner in verschiedenen Orten mit 
unterschiedlichen Betriebssystemen (NJE = Network Job Entry).Es können sowohl Jobs als auch 
Output in jede Richtung geschickt werden. Heute wird dieser Weg häufig als Datentransfer miss-
braucht. Voraussetzung ist JES2 oder JES3 als Job-Management-System. 

7.1.7.12. Purge-Phase 

Alle Bereiche, die von dem betreffenden Job innerhalb der Spooldatei belegt wurden, werden frei-
gegeben. Die Eintragungen der beendeten Jobs werden aus der Job-Output-Queue gelöscht. Der 
Operator wird über das Ende der Purge-Phase informiert. 

7.2. Steueranweisungen 

Generell gilt: Alle JCL-Anweisungen müssen in Upper Case kodiert sein. In der Job Control Langu-
age gibt es eine Reihe unterschiedlicher Anweisungstypen. Diese sind: 

Anweisungstyp Bedeutung 

JOB Gibt dem Arbeitsauftrag einen Namen 

EXEC Ruft ein Programm oder eine JCL-Prozedur auf 

DD Erklärt das Datenuniversum für ein Programm 

NULL Früher zwingend als ENDE, wird heute nicht mehr gebraucht 

JCLLIB Zuweisung privater Procedure-Bibliotheken 

SET Zuweisung von Variablen 

INCLUDE Import von JCL vergleichbar zu File Tailoring 

JES-Command Ausführung von JES-Commands via JCL. Ist nur wenigen Mitarbeitern 
erlaubt. 

JES-Anweisungen Routing, Drucker-/Checkpoint-Definitionen 

PROC Procedure-Beginn-Anweisung 

PEND Procedure-END-Anweisung 

 

Viele Verarbeitungen lassen sich jedoch mit zwei dieser Typen erledigen. Die Minimalangaben sind 
die Anweisung EXEC und deren dazugehörige DD-Statements. 

7.2.1. Reihenfolge der Anweisungen 

//name    JOB ..... 

//        PROC .... 

//        .... 

//        PEND 

//JOBLIB  DD ...... 

//name    EXEC .... 

//STEPLIB DD ...... 

//name    DD ...... 

//name    DD DATA,DLM=## 

daten 

## 

//name    DD ...... 

// 

7.2.2. Aufbau und Einteilung der Anweisungen 

Die JCL-Statements unterteilen sich in verschiedene Felder. Obligatorisch und Erkennungszeichen 
der JCL-Befehle sind Schrägstriche(//) auf den Spalten 1 und 2. 

Abgelegt werden die Statements in einer Datei fester Länge von 80 Byte (ein Relikt aus der Zeit der 
Lochkarten). 

Es gibt insgesamt 4 Felder. NAME, TYP, OPERAND(EN), KOMMENTAR. Zwischen den einzelnen 
Feldern muss mindestens eine Leerstelle sein. 

Das Namensfeld muss auf Spalte 3 beginnen und darf maximal 8 Byte (alphanumerisch) lang sein. 
Die erste Stelle muss ein Buchstabe sein. 

In der JOB-Anweisung muss das Namensfeld angegeben sein. 

In der EXEC-Anweisung muss das Namensfeld zwingend angegeben werden, wenn an anderer 
Stelle auf diesen Step Bezug genommen werden soll. Der besseren Transparenz wegen sollte es 
aber immer versorgt werden. 

In der DD-Anweisung wird der symbolische Dateiname angegeben, welcher im Programm benutzt 
wird. Der Name kann nur bei Dateiverkettung ab der zweiten Stelle weggelassen werden. 

Im Operandenfeld stehen Anweisungen für die Operation. Wenn die Operanden nicht in einer Zeile 
Platz finden, kann in der nächsten Zeile fortgefahren werden. Die Trennung kann nur nach vollstän-
diger Angabe eines Operanden erfolgen und wird durch Abschluss einer Zeile mit Komma(,) durch-
geführt. 

Die Fortsetzungszeile beginnt mit //. Die Operanden müssen dann zwischen den Spalten 4 und 16 
weitergeschrieben werden. 

Das Kommentarfeld muss durch eine Leerstelle vom Operandenfeld getrennt werden. 

In der Fortsetzungsspalte(72) kann ein Zeichen ungleich BLANK angegeben werden. 

7.2.3. Bedeutung der Sonderzeichen 

Name Zeichen Bedeutung 

Alphabetic 
Numeric 
At 
Hash 
Dollar Sign 

A-Z 
0-9 
@ 
# 
$ 

Namen der Anweisungen, Stelle 1 muss ein Buchstabe sein 

Comma , Trennzeichen zwischen Parametern 

Period . Trennung der Dateinamensteile 

Slash / Merkmal für JCL auf den Stellen 1 und 2 

Apostrophe ' Schließt alle Angaben eines Parameters ein und erlaubt dabei Son-
derzeichen 

Parenthesis () Eingrenzung von Sub-Parametern 

Asterisk * Referenzangaben, meist Rückbezugnahmen 

Plus 
Hyphen 

+ 
- 

Kennzeichnet die relative Generation einer Generation Data Group 

Equal = Wertzuweisung für Schlüsselworte 

Blank  Feldbegrenzer 

 

Werden andere als die Genannten oder generell erlaubte Zeichen am falschen Ort kodiert, führt dies 
immer zum Abbruch von JCL. 

Es ist empfehlenswert, sich die englischen Namen der Sonderzeichen einzuprägen. In einem Fehler-
protokoll steht womöglich "Incorrect use of parenthesis" aber mit Sicherheit niemals "Du habst die 
Klammern falsch gesetzt". 

7.3. Die Job-Anweisung 

//jobname JOB operanden            

7.3.1. Jobname 

Mit dem Jobnamen wird ein Auftrag im System definiert. Die Angabe ist zwingend. 

7.3.2. Positionsparameter oder Stellungsparameter 

Positionsparameter müssen immer vor den Schlüsselwortparametern geschrieben werden. Die Rei-
henfolge der Stellungsparameter ist einzuhalten. Zwei Angaben sind möglich. Wird die erste Angabe 
weggelassen, muss ein Komma(,) gesetzt werden. 

7.3.2.1. account-information 

Der erste Stellungsparameter ist die Abrechnungsinformation. Die Angaben werden zwischen runde 
Klammern() gesetzt. Wird mehr als eine Angabe gemacht, sind die einzelnen Angaben innerhalb der 
Klammern durch Komma zu trennen. 

7.3.2.2. programmierer-name 

ist der zweite Stellungsparameter. Hier kann eine beliebige Eintragung bis zu 20 Byte Länge ange-
geben werden. Um Probleme bei der Verwendung von Sonderzeichen auszuschließen, sollte dieser 
Eintrag zwischen Hochkommata gesetzt werden. 

Beispiel 

//SP11 JOB (11199002),'AUSBILDUNG', Stellungsparameter 

//           CLASS=...                Schlüsselwort  

7.3.3. Schlüsselwort-Parameter 

Die Reihenfolge der Schlüsselwortparameter ist beliebig, da sie auf Grund ihres Namens erkannt 
werden. Wird ein Parameter nicht angegeben, so ist, soweit definiert, der Standardwert gültig. 

Mögliche Angaben (Defaultwerte sind angegeben. Diese sind aber abhängig von der jeweiligen In-
stallation und nicht zwingend so): 

Schlüsselwort Bedeutung 

CLASS Verarbeitungsklasse, in welcher der JOB läuft 

COND Regelt die Ablaufbedingung(en) unter denen eine Verarbeitung stattfinden soll 
(oder nicht). 

MSGCLASS Benennt die Ausgabeklasse, in der das Verarbeitungsprotokoll abgelegt wird. 

MSGLEVEL Bestimmt, wie aussagekräftig und umfangreich das Verarbeitungsprotokoll ist. 

NOTIFY Benennt einen Benutzer, der bei Jobende benachrichtigt wird. 

REGION Bestimmt die Adressraumgröße für die Verarbeitung 

TIME Kann eine Laufzeitbegrenzung durchführen 

ADDRSPC Entscheidet, ob Pageing für die Verarbeitung zulässig ist 

PERFORM Bestimmt eine Performance Gruppe (wird heute meist durch WLM-Angaben 
geregelt) 

PRTY Bestimmt die Auswahlpriorität im Spoolsystem 

7.4. Die EXEC-Anweisung 

//stepname EXEC┬PROC=prozedurename┬───────────────────►◄ 

               └PGM=programmname──┘ 

Die EXEC-Anweisung wird benutzt, um festzulegen, welches Programm aufzurufen ist, bzw. welche 
JCL-Prozedur an Stelle der Anweisung eingefügt werden soll. 

7.4.1. Stepname 

Ist nicht zwingend erforderlich, außer es wird auf den Step Bezug genommen, sollte aber angegeben 
werden. Anderenfalls ist keine Referenzangabe auf Stepebene möglich. 

7.4.2. Stellungsparameter 

Als Stellungsparameter kann der Programm- oder Prozeduraufruf angegeben werden. 

7.4.2.1. Programmaufruf: 

//name EXEC PGM=programmname 

programmname muss Member einer gültigen Lademodul-Bibliothek sein. Liegt das Programm in 
keiner standardgemäß zugeordneten Lademodulbibliothek, muss die Ladedatei durch eine JOBLIB- 
oder STEPLIB-DD-Anweisung zugeordnet werden. 

7.4.2.2. Prozeduraufruf: 

//name EXEC PROC=prozedurname  

prozedurname ist Member einer JCL-Prozedurbibliothek, oder der in der PROC-Anweisung ange-
gebene Name einer Instream-Prozedur. 

7.4.3. Schlüsselwort-Parameter 

Schlüsselwort Bedeutung 

ACCT 
ADDRSPC 
COND 
REGION 
PERFORM 
TIME 

Siehe Job-Anweisung 

DYNAMNBR Maximale Anzahl der in einem Programm dynamisch zuweisbaren Dateien 
(nicht über DD-Anweisung) 

7.5. Die DD-Anweisung (Daten-Definition) 

//ddname DD operanden 

Sie beschreibt eine Datei in einem Job und definiert die Ein-/Ausgabeeinheit für diese Datei. Außer-
dem wird die Verbindung zum symbolischen Namen der Datei im Programm hergestellt. Maximal 
können 1635 DD-Anweisungen (Größe der Task Input Table 32KB) oder 3273 (TIOT = 64KB) in 
einem Job liegen. 

Um eine DD-Anweisung in einer Prozedur zu verändern, müssen folgende Punkte berücksichtigt 
werden: 

• Vor den DD-Namen muss der Stepname geschrieben werden. Beispiel (//STEP1.EING1 DD 
...).  

• Es müssen nur die Operanden angegeben werden, die verändert werden sollen.  

• Wenn mehr als eine DD-Anweisung geändert werden soll, muss die Reihenfolge mit der Rei-
henfolge in der Prozedur übereinstimmen.  

• Wird ein DD-Name verwendet, der in der Prozedur nicht vorkommt, wird die Anweisung hin-
zugefügt. 

7.5.1. DD-Name 

Benennt die Anweisung. DD-Namen in einem Step müssen eindeutig sein. 

Wird ein DD-Name weggelassen, wird die Datei mit dem zuletzt angegebenen DD-Namen verkettet. 

Folgende DD-Namen dürfen nicht beliebig benutzt werden 

DD-Namen Bedeutung 

JOBLIB Private Bibliothek für PGM= 

STEPLIB 

SYSMDUMP Dump-Ausgaben 

SYSUDUMP 

SYSABEND 

JOBCAT Sonstige 

SYSCKEOV 

SYSCHK 

IMAGELIB 

JESJCL Reserviert für JES 

JESJCLIN 

JESMSGLG 

JESYSMSG 

7.5.2. Stellungsparameter 

 
Nachfolgende Parameter können nur alternativ angegeben werden. 

Stellungsparame-
ter 

Bedeutung 

* Der Beginn von Instream-Daten wird definiert, die unmittelbar dahinter fol-
gen. Das Ende der Daten kann durch eine Anweisung /* gekennzeichnet 
werden. 

DATA, DLM=xx DATA wird benutzt, wenn Daten auf den Stellen 1 und 2 // aufweisen. Dies 
ist der Fall, wenn JCL-Statements als Instream-Daten verarbeitet werden. 
Das Ende der Daten wird durch die Anweisung /* gekennzeichnet. 
DLM= definiert ein Begrenzungszeichen (2 Byte lang). 

DUMMY Schaltet eine Datei aus. Die übrigen Operanden können im Anschluss an 
DUMMY geschrieben werden. Sie werden auf syntaktische Fehler geprüft, 
ansonsten aber ignoriert. 

DSN=NULLFILE ist gleichbedeutend mit DUMMY. 

Wenn im DCB des Programms keine BLKSIZE beschrieben ist, muss die 
Angabe in der DD-Anweisung erfolgen, da sonst ein Fehler auftritt. 

 

Beispiel: 

//AUSGABE DD DUMMY,DCB=BLKSIZE=132 

7.5.3. Schlüsselwortparameter 

Wie bereits erwähnt, können Schlüsselworte in beliebiger Reihenfolge kodiert werden, da sie auf-
grund ihrer Namen erkannt werden. Wir wollen hier nicht alle exotischen Operanden anführen, son-
dern nur diejenigen, welche im Alltagsleben am häufigsten gebraucht werden (der Schwerpunkt des 
Buches ist REXX). Die Details zu den jeweiligen Operanden können im Referenzteil (Band2) nach-
gelesen werden. 

Schlüsselwort Bedeutung 

DSN Dateiname für die Verarbeitung 

DISP Definiert, ob die Datei vorhanden ist, oder nicht, die Art des Zugriffes (ex-
klusiv oder nicht) und was bei Ende der Verarbeitung im Normalfall (normal 
END – 2. Parameter) sowie im Falle eines Abbruchs (ABEND – 3. Para-
meter) passieren soll.) 

UNIT Gerätegruppenangabe, im Unternehmen nachzufragen. Allgemein gültige 
Angaben wie DASD (Direct Access Storage Device – Platte, sind nicht er-
wünscht. 

SPACE Platz-Größenangaben für neu zu erstellende Dateien auf Platte 

DCB Data Control Block, beschreibt den physischen Aufbau einer Datei. Meist 
nur nötig bei Neuanlage. 

LABEL Nur bei Bandverarbeitung. Liefert meist Angaben für Kennsätze, die bei 
der Verarbeitung von Kundendatenträgern benötigt werden, oder definiert 
Sperrfristen (Verfalldatum) 

LIKE Referenz für DCB und SPACE bei SMS-managed-Datasets 

MGMTCLAS 
STORCLAS 
DATACLAS 

Definitionen für das Storage Management Subsystem (SMS) für Manage-
mentclass, Storageclass und Dataclass. Hier sollte man wissen, was man 
tut (gegebenenfalls hilft eine Rückfrage bei den Storage-Managern) 

 

7.5.4. Das Zusammenwirken 

Die eigentliche Schwierigkeit bei der Erstellung von Job Control Abläufen liegt nicht an der JCL an 
sich, sondern in einem detaillierten Wissen um die Arbeitsweise der Programme, für welche die JCL 
erstellt werden soll. 

Um dem gerecht zu werden müsste im Grunde jede Open-Instruktion, jedes Select-Statement in 
einem (Cobol)Programm gekannt werden und auch, welche Verarbeitung das Programm unter dem 
jeweiligen Synonym durchführt. Nur dann kann sinnvoll entschieden werden, wie die dazu passende 
Job Control auszusehen hat. 

Wir wollen hier anhand einiger Beispiele zeigen, wie die Parameter der JCL zusammenspielen. Dazu 
benutzen wir einige der gängigen Utilities (IEBGENER und DFSORT). 

7.6. Simple Beispiele 

7.6.1. IEBGENER 

IEBGENER ist eines der sehr häufig benutzten Utilities in z/OS mit klassischer Ein-/Ausgabeverar-
beitung. Sofern gewollt, können die Eingabedaten auf dem Weg zur Ausgabe manipuliert werden. 

Sehen wir uns zunächst das Funktionsprinzip des IEBGENER an. 

Generell kann aus diesem Schaubild bereits vieles über die zu erstellende JCL abgelesen werden. 
Das JCL-Gerippe wird folgendermaßen aussehen: 

//stepname EXEC PGM=IEBGENER 

//SYSUT1   DD hier wird die Eingabe beschrieben 

//SYSUT2   DD hier wird die Ausgabe beschrieben 

//SYSIN    DD hier liegen die Steuerbefehle des IEBGENER 

//SYSPRINT DD das ist die Protokolldatei der Verarbeitung 

Unterstellen wir folgende Verarbeitung: 

Das Member ADR1 der Datei SP1.EDU.DATA soll eingelesen und zur Druckausgabe über die 
Outputklasse A gebracht werden. 

//DRUCK    EXEC PGM=IEBGENER 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA(ADR1) 

//SYSUT2   DD SYSOUT=A 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

Die Überlegung, die DD-Statements für SYSIN und SYSPRINT ganz einfach wegzulassen, wenn 
man keine Angaben dazu kodieren will, ist verständlich. Leider führt dies aber zum Abbruch der 
Verarbeitung. Warum? IEBGENER geht nicht davon aus, dass das das Fehlen der Statements be-
deutet, dass weder Steueranweisungen gelesen (SYSIN) noch ein Verarbeitungsprotokoll ausgege-
ben (SYSPRINT) werden soll. Um zu erkennen, ob und welche Steueranweisungen da sind, muss 
IEBGENER diese lesen und um das zu können, muss der symbolische Dateiname SYSIN geöffnet 
werden. Da dieser aber nicht zugewiesen ist, führt dies zu einer Abbruchmeldung "SYSIN-DD-State-
ment is missing". Gleiches gilt für SYSPRINT. Auch hier müsste vor dem WRITE ein OPEN passie-
ren. Dass es unter SYSIN keine Sätze zu lesen gibt und die Protokolldaten für SYSPRINT in den 
Wind zu schreiben sind, regelt die Angabe DUMMY. 

Nun wollen wir das Member ADR1 lesen und mit diesen Daten eine sequentielle Datei erstellen. 
Folglich brauchen wir unter SYSUT2, der Datenausgabe des IEBGENER, alle Angaben, welche zum 
Erstellen eines sequentiellen Datasets nötig sind. Hierzu zählen: 

▪ Neuanlage 

▪ Katalogisieren 

▪ Größenangaben 

▪ Dateiformat 

//COPY1    EXEC PGM=IEBGENER 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA(ADR1) 

//SYSUT2   DD DSN=SP1.DEMOSEQ.DATA, 

//            DISP=(NEW,CATLG,DELETE), 

//            UNIT=SYSDA, 

//            SPACE=(TRK,(5,5)), 

//            DCB=(LRECL=132,BLKSIZE=0,RECFM=FB) 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

Die Angabe UNIT= sollte im jeweiligen Unternehmen erfragt werden. SYSDA ist in der IBM-Welt 
eine pauschale Aussage für Platten. In der Regel werden unter Einsatz des SMS (Storage Manage-

ment Subsystem) vom jeweiligen Betreuer sog. Unit-Gruppen erstellt, die eine Vielzahl von Informa-
tionen enthält. Eine Abweichung von den Unitgruppen des Unternehmens kann zur Folge haben, 
dass derartige Dateien ohne Rückfragen gelöscht werden. 

Nehmen wir ein anderes Beispiel das Member ADR1 der SP1.EDU.DATA soll als Member ADRESS 
der Datei SP1.DEMO.DATA weggeschrieben werden. Die Datei soll im Zuge der Verarbeitung er-
stellt werden. Die Kenngrößen und der Aufbau der neuen Datei sollen denen der alten entsprechen. 
Wir unterstellen, dass SMS im Einsatz ist (ich kenne heute kein Unternehmen mehr, das z/OS ohne 
SMS einsetzt). 

//COPY2    EXEC PGM=IEBGENER 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA(ADR1) 

//SYSUT2   DD DSN=SP1.DEMO.DATA(ADRESS), 

//            DISP=(NEW,CATLG,DELETE), 

//            LIKE=SP1.EDU.DATA 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

Das Beispiel oben wartet mit einer "lästigen" Konstellation auf. Die Ausgabedatei unter SYSUT2 soll 
mit DISP=(NEW,CATLG) erstellt und bei Programmende katalogisiert werden. Das kann aber gene-
rell nur einmal funktionieren. Danach ist die Datei vorhanden und ein zweiter SUBMIT würde zu 
einem JCL-Error führen. Dementsprechend müsste man die Datei vor erneuter Verarbeitung löschen 
oder die Disposition auf OLD ändern. 

Die ganz Schlauen begegnen diesem Problem mit einem vorgeschalteten Step, bei dem die Datei 
gelöscht werden soll. Ist sie nicht vorhanden, würde das im IDCAMS einen entsprechenden RC 
erzeugen, den man aber ignorieren könnte. 

//DELETE   EXEC PGM=IDCAMS 

//SYSPRINT DD SYSOUT=* 

//SYSIN    DD * 

  DELETE SP1.DEMO.DATA  

  SET MAXCC=0 

//COPY2    EXEC PGM=IEBGENER 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA(ADR1) 

//SYSUT2   DD DSN=SP1.DEMO.DATA(ADRESS), 

//            DISP=(NEW,CATLG,DELETE), 

//            LIKE=SP1.EDU.DATA 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

Die wirklich beste Lösung, wäre aber, ein Programm einzubinden, das in allen Installationen verfüg-
bar ist, in denen IWS (vormals OPC) im Einsatz ist: das EQQDELDS. Dieses Programm untersucht 
die der Verarbeitung zugrunde liegende JCL auf DISP-Angaben NEW und prüft, ob diese Dateien 
bereits vorhanden sind. Wenn ja, werden sie gelöscht. Generell benötigt das EQQDELDS keine DD-
Statements. Wenn aber ein SYSPRINT-DD-Statement kodiert ist, schreibt das Programm dorthin, 
was es getan hat. 

//DELETE   EXEC PGM=EQQDELDS 

//SYSPRINT DD SYSOUT=* 

//COPY2    EXEC PGM=IEBGENER 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA(ADR1) 

//SYSUT2   DD DSN=SP1.DEMO.DATA(ADRESS), 

//            DISP=(NEW,CATLG,DELETE), 

//            LIKE=SP1.EDU.DATA 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

Bei Mitarbeitern, welche nichts oder nur wenig mit IWS zu tun haben, ist diese Lösung weitgehend 
unbekannt. Schade. 

7.6.2. DFSORT 

Sehen wir uns ein weiteres der Utilities an: DFSORT. Der Sort stellt wohl eines der vielseitigsten 
und auch mächtigsten Dienstprogramme dar. Bis auf wenige Ausnahmen kann der Sort für nahezu 
alles benutzt werden, was das Alltagsleben so bringt. Ich persönlich setze ihn zu mehr als 80% 
meiner Standards ein. Der Name beschreibt ihn nur unzureichend. Sort kann auch normale sequen-
tielle Ein-/Ausgaben durchführen oder bietet auch eine Merge-Funktion. Zunächst das Funktions-
schaubild: 

 

Auch hier kann im Wesentlichen bereits erkannt werden, was alles kodiert werden muss. Das JCL-
Gerippe wird folgendermaßen aussehen: 

//stepname EXEC PGM=SORT 

//SORTIN   DD die unsortierte Eingabe 

//SORTOUT  DD die Ausgabe 

//SORTWK01 DD Arbeitsbereich 

//SORTWKnn DD bis zu 16 können definiert werden 

//SYSOUT   DD das Arbeitsprotokoll 

//SYSIN    DD die Steuerbefehle sind zwingend 

Sehen wir uns im Folgenden an, wie eine sequentielle Ein-/Ausgabe unter Zuhilfenahme des Sort 
aussieht: 

//COPY3    EXEC PGM=SORT 

//SORTIN   DD DSN=SP1.EDU.DATA(ADR1),DISP=SHR 

//SORTOUT  DD SYSOUT=* 

//SYSOUT   DD SYSOUT=* 

//SYSIN    DD * 

  SORT FIELDS=(COPY) 

Die Angabe der Workbereiche ist optional. Generell ist der Sort so intelligent, den zugewiesenen 
Arbeitsspeicher zu benutzen. Sofern dieser für die Sortierung genügt, brauchen keine SORTWK-
Bereiche angegeben werden. 

Nun versuchen wir eine etwas komplexere Aufgabenstellung zu lösen: 

Aufbau der Eingabedaten: 

----+----1----+----2----+----3---Anschrift--2----+----3—- 

Vorname        Nachname                      tt.mm.jjjj 

Vorname von Stelle 1 bis 15  
Nachname von Stelle 16 bis 33  
Anschrift und sonstiges von Stelle 34 bis 120  
Geburtsdatum von Stelle 121 bis 130  
Stelle 131 bis 132 leer 

Um nebenbei die Leistungsfähig des Sort aufzuzeigen, soll folgendes gelöst werden: 

Eingabe ist das Member ADR1 der SP1.EDU.DATA. 

Als Ausgabe soll eine Liste erstellt werden, welche in die Output-Hold-Queue des Spoolsystems 
gestellt wird. 

Die Ausgabesätze sollen von links nach rechts das Geburtsdatum, den Nachnamen und den Vor-
namen enthalten. 

Um möglichst schnell sehen zu können, wo es heute Freibier gibt, soll die Sortierung vom 1. Bis zum 
letzten des Monats laufen. Sind mindestens zwei am gleichen Tag geboren, sollen die Jüngeren vor 
den Älteren liegen. Ist bei mindestens 2 das Geburtsdatum identisch, soll nach Nachnamen sortiert 
werden. 

Damit die Ausgabe aktuell bleibt, sollen nur diejenigen Sätze enthalten sein, welche den Geburts-
monat Januar haben. 

Damit auch dies mal zu sehen ist, wird eine JOB-Anweisung mitgeliefert. Hier die entsprechende 
Job Control: 

//SP11   JOB (11199002),'GEBLISTE', Account,Info 

//            CLASS=A,                Verarbeitungsklasse 

//            MSGCLASS=X,             JobLog Protokoll 

//            NOTIFY=SP1              Nachricht Jobende 

//************************************************ 

//* Erstellen Freibier-Liste 

//GEBLISTE EXEC PGM=SORT 

//SORTIN   DD DSN=SP1.EDU.DATA(ADR1),DISP=SHR 

//SORTOUT  DD SYSOUT=* 

//SORTWK01 DD UNIT=VIO,SPACE=(CYL,(5,5)) 

//SORTWK02 DD UNIT=VIO,SPACE=(CYL,(5,5)) 

//SORTWK03 DD UNIT=VIO,SPACE=(CYL,(5,5)) 

//SORTWK04 DD UNIT=VIO,SPACE=(CYL,(5,5)) 

//SYSOUT   DD SYSOUT=* 

//SYSIN    DD * 

 SORT FIELDS=(124,2,ZD,A,121,2,ZD,A,127,4,ZD,D,16,5,CH,A) 

 INCLUDE COND=(124,2,EQ,C'01'),FORMAT=CH 

 OUTREC FIELDS=(121,10,C' ',16,18,1,15) 

Die JOB-Anweisung vergibt den Namen für den Arbeitsauftrag.  

Die EXEC-Anweisung vergibt den Namen für den Step und legt fest, wie das zu rufende Programm 
heißt. Da es bei SORT sehr viele Varianten von verschiedenen Anbietern gibt, wird der jeweils ein-
gesetzte SORT mit einem ALIAS versehen, damit er, wie er auch immer heißen mag, über den 
Namen SORT aufgerufen werden kann. 

Die SORTIN-DD-Anweisung beschreibt die Eingabedatei.  Als Input wird das Member ADR1 der 
Datei SP1.EDU.DATA festgelegt. Die Art des Zugriffs ist SHR (Mehrfachzugriffe während der Ver-
arbeitung sind erlaubt). Dies ist sinnvoll, weil die Verarbeitung nur lesen will. 

SORTOUT legt fest, wohin der sortierte Output geschrieben wird. Hier sind die Output-Bereiche des 
Spools angegeben. Mit Stern erfolgt eine Rückbezugnahme auf die Angabe MSGCLASS= der 
Jobanweisung (Klasse X). Ist keine Jobanweisung kodiert oder dort keine MSGCLASS-Angabe ent-
halten wird der Default benutzt (in der Regel die Klasse X). 

Die Sortwork-Bereiche werden im Arbeitsspeicher gelegt (UNIT=VIO = Virtual I/O).  Jeder Workbe-
reich bekommt 5 Cylinders primary Space und kann um jeweils 5 Cylinders vergrößert werden, bis 
das Maximum an Extents erschöpft ist. 

Das Arbeitsprotokoll des Sort (SYSOUT) wird zusammen mit dem JobLog und den Ausgabedaten 
im Spoolausgabebereich abgelegt. 

Die Steueranweisungen (SYSIN) werden Instream kodiert und folgen im Anschluss. 

Die SORT-FILED-Anweisung regelt die Sortierfolge. Immer vier Angaben beschreiben ein Feld 
durch Beginnposition, Feldlänge, Feldformat und Sortierrichtung. Die INCLUDE-Anweisung be-
stimmt, wie Datensätze in der Eingabedatei aussehen müssen, um verarbeitet zu werden. 

Die OUTREC-Anweisung steuert den Aufbau des Ausgabedatensatzes. Je Feld wird eine Beginn-
position und eine Länge bezogen auf den Eingabedatensatz gemacht. Die auf diese Weise beschrie-
benen Felder werden im Ausgabesatz von links nach rechts zusammengefügt. 

7.7. Temporäre Dateien 

Oft ist es so, dass zur Gesamtlösung eines Problems mehr als ein Programm benötigt wird. Das 
erste übernimmt eine Teilaufgabe und stellt das Ergebnis in einer temporären Datei für den Folge-
step zur Verfügung. Der zweite Step erzeugt das Endergebnis und muss dazu den Namen der Zwi-
schendatei nicht kennen. Auf diesen Namen wird mittels einer Back-Reference gezeigt: 

//SP1A     JOB (4711),CLASS=A,NOTIFY=&SYSUID 

//* Demo für den Umgang mit TEMP-Dateien, welche in einem Folgestep 

//* noch verarbeitet werden sollen. 

//* Generell wäre die Verarbeitung natürlich ein EINEM Step möglich 

//SORT1    EXEC PGM=SORT 

//SORTIN   DD DISP=SHR,DSN=SP1.EDU.DATA(ADRTEST) 

//SORTOUT  DD DISP=(NEW,PASS),STORCLAS=ALTEMP,SPACE=(CYL,1) TEMP-OUTPUT 

//SYSOUT   DD SYSOUT=* 

//SYSIN    DD * 

  SORT FIELDS=(84,2,ZD,A,81,2,ZD,A,87,4,ZD,A,1,5,CH,A) 

  INCLUDE COND=(84,2,EQ,C'01'),FORMAT=CH 

/* 

//SORT2    EXEC PGM=SORT 

//SORTIN   DD DISP=OLD,DSN=*.SORT1.SORTOUT                  TEMP-INPUT 

//SORTOUT  DD SYSOUT=* 

//SYSOUT   DD SYSOUT=* 

//SYSIN    DD * 

  SORT FIELDS=(COPY) 

  OUTREC FIELDS=(80,10,C' ',1,29) 

/* 

7.8. Variablen in der JCL 

Generell können in einer JCL auch Variablen (oder auch die sogenannten System Symbols) benutzt 
werden. Welche das im Detail sind, ist installationsabhängig. Es gibt eine Reihe von Symbols, die 
IBM-Standard sind. Einige davon sind statisch, andere dagegen dynamisch. Das heißt, dass sie sich 
im laufenden Betrieb verändern können (wie Datum oder Uhrzeit). Darüber hinaus können beliebige 
firmenspezifische Symbols definiert werden, welche unter Umständen auf jeder LPAR andere sind 
oder bei gleichnamigen unterschiedliche Werte enthalten können. 

Welche Symbole es gibt und wie deren aktuellen Werte sind, kann im SDSF über das Symbol-Panel 
angezeigt werden (sofern man es aufrufen darf). Bei den Beispielprogrammen für ISFSLASH ist 
eines dabei, welches diese Symbole ermittelt und zusammen mit ihren Werten anzeigt. 

Grundsätzlich können Variablen auch direkt in der JCL definiert werden. Dies ist beispielsweise dann 
sehr sinnvoll, wenn bestimmte Dinge (z.B.: Dateinamen oder DISP-Parameter) mehrfach angege-
ben werden sollen und nicht immer gleichnamig kodiert werden müssen. Das macht eine JCL pfle-
geleichter und weniger fehleranfällig. 

Um Systemsymbole in der JCL benutzen zu können, ist Voraussetzung, dass der Parameter SYS-
SYM=ALLOW in der jeweiligen Jobklasse definiert ist. 

Symbole werden generell mit einem führenden Ampersandzeichen(&) kodiert. 

JCL-Symbole sind für gewöhnlich Variablendefinitionen, welche im PROC-Statement oder über 
SET-Anweisungen definiert werden. 

Soll der Inhalt eines Symbols direkt mit einem String verkettet werden, ist zwischen dem Symbol-
namen und dem String ein Punkt(.) zu kodieren. Dies ist notwendig, damit erkannt werden kann, 
wo der Symbolname endet und der String beginnt 
 

7.8.1. Variablen per SET-Befehl 

//SP1A JOB (4711),'WITTEMANN',CLASS=A,NOTIFY=SYSUID 

//* Variablendefinition 

//MYDSN SET MYDSN=SP1.RESULT.DATA 

//DISP1 SET DISP2=(NEW,CATLG) 

//DISP2 SET DISP1=SHR 

//* Verarbeitung 

//GENER1   EXEC PGM=IEBGENER 

//SYSUT1   DD DISP=SHR,DSN=hier_steht_die_eingabedatei 

//SYSUT2   DD DISP=&DISP1,DSN=&MYDSN 

//SYSIN    DD * 

  Steuerbefehle für den IEBGENER 

//SYSPRINT DD SYSOUT=* 

//SORT     EXEC PGM=SORT 

//SORTIN   DD DISP=&DISP2,DSN=&MYDSN 

//SORTOUT  DD ... 

: 



Inside z/OS Seite 392 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

Selbstverständlich könnte man so etwas auch mit einer Back-Referenz erledigen, aber es geht ja 
nur darum, die Funktionsweise zu erkennen. 

7.8.2. System Symbols 

//CREDSN   EXEC PGM=IEFBR14                                             

//MYDSN    DD DSN=&SYSUID..D&YYMMDD..T&HHMMSS..DATA,SPACE=(TRK,1) 

7.8.3. Variablen auch in Instream-Daten 

Seit einiger Zeit ist es möglich, Variablen auch innerhalb von Instreamdaten zu benutzen. Möglich 
wird dies durch das Zusammenspiel des EXPORT-Statements und SYMBOLS-Parameters in einer 
DD-Anweisung. Über letzteren kann auch gesteuert werden, ob das auführende oder das einlesende 
MVS für die Substituierung der Variablen zuständig sein soll. 

//SP1A     JOB (4711),CLASS=A,MSGCLASS=H,NOTIFY=&SYSUID 

//EXPORT1  EXPORT SYMLIST=(SYSUID,SYSNAME) 

//TSO1     EXEC PGM=IKJEFT01 

//SYSTSPRT DD SYSOUT=* 

//SYSTSIN  DD *,SYMBOLS=CNVTSYS 

  EX '&SYSUID..P.EXEC(SAGWAS)' 'TRALLALA' EXEC 

  LISTD '&SYSUID..&SYSNAME..ISPF.PROFILE' ST H 

/* 

7.8.4. Instream-Variablen mit unterschiedlicher Auflösung 

Hier gehen wir davon aus, dass der SUBMIT auf MVS1 passiert, die Verarbeitung dagegen auf 
MVS2. Beide Systeme laufen auf dem gleichen Plex und greifen per Shared Dasd auch die gleichen 
Dateien zu. 

//        JOB (RZRZ),'HOSNFURZER',CLASS=A,MSGCLASS=H, 

//            NOTIFY=&SYSUID,SCHENV=MVS2 

//MYEXPR1  EXPORT SYMLIST=(USER,MVS) 

// SET USER=&SYSUID 

// SET MVS=&SYSNAME 

//TSO1  EXEC PGM=IKJEFT01 

//SYSTSPRT DD SYSOUT=* 

//SYSTSIN  DD *,SYMBOLS=CNVTSYS 

  listd '&user..&mvs..ispf.profile' m 

//* 

//TSO2  EXEC PGM=IKJEFT01 

//SYSTSPRT DD SYSOUT=* 

//SYSTSIN  DD *,SYMBOLS=EXECSYS 

  listd '&user..&sysname..ispf.profile' m 

7.9. Verarbeitung von Skeletten 

Wirklich schlimm ist redundanter Code. Nicht nur, dass bei Änderungen sehr viel Aufwand betrieben 
werden muss um nichts zu vergessen, durch viele Redundanzen wird alles sehr fehleranfällig. 

Wir wollen uns an einem kleinen Beispiel ansehen, wie man dem entgegenwirken kann und nutzen 
dazu TSO im Batch. 

In den meisten Fällen wird in diese JCL neben den privaten Bibliotheken auch die Umgebung des 
Unternehmens in Punkto SYSPROC/SYSEXEC integriert. Auf diese Weise ist die Definition der fir-
menspezifischen Dateien hunderte Male in irgendwelchen Jobs enthalten. Und wenn sich daran 
etwas ändert? Dann ist das Geschrei riesig und der Änderungsaufwand oft nicht zu beziffern. Und 
wird ohne Vorankündigung an der firmenspezifischen Umgebung etwas geändert, ist das Drama 
groß. Neu hinzugefügte Dateien fehlen, Nicht mehr existente Dateien führen zu einer Flut von JCL-
Fehlern. 

Betrachten wir zunächst die JCL, welche wir für TSO im Batch benötigen: 

//TSOBAT EXEC PGM=IKJEFT01   Aufruf des TSO 

//SYSTSIN  DD *                               Tastatur 

  %MYREXX 

//SYSTSOUT DD SYSOUT=*                        Bildschirm 

//SYSEXEC  DD DISP=SHR,DSN=SP1.MYREXX.EXEC    Meine REXXe 

Solange alle Subroutinen, die aus MYREXX gerufen werden, in der gleichen Bibliothek zu finden 
sind, ist die Welt in Ordnung. Wie bindet man jetzt aber die Tool-Umgebung des Unternehmens ein? 
Hierzu erst ein schlechtes Beispiel, wie es leider sehr häufig anzutreffen ist: 

//TSOBAT EXEC PGM=IKJEFT01   Aufruf des TSO 

//SYSTSIN  DD *                               Tastatur 

  %MYREXX 

//SYSTSOUT DD SYSOUT=*                        Bildschirm 

//SYSEXEC  DD DISP=SHR,DSN=SP1.MYREXX.EXEC    Meine REXXe 

//         DD DISP=SHR,DSN=SYS1.ISPF.SISPEXEC 

//         DD DISP=SHR,DSN=SYS1.SBPXEXEC 

//         DD DISP=SHR,DSN=ABC.COMPANY.ALLG.EXEC 

//         DD DISP=SHR,DSN=ABC.COMPANY.TOOLS.EXEC 

//         DD DISP=SHR,DSN=ABC.COMPANY.FUNC.EXEC 

Der Nachteil hierbei ist, dass es bei Änderungen der firmenspezifischen Bibliotheken zu Problemen 
kommt, sofern neue Dateien hinzukommen (weiß doch keiner) oder wenn welche wegkommen (JCL-
Error wegen "Dataset not found"). Die Lösung hierfür wäre, dass die zentralen Bibliotheken in sepa-
raten Membern abgelegt werden, welche per INCLUDE-Anweisung angezogen werden können. Auf 
diese Weise können sie an zentraler Stelle gepflegt werden, der Pflegeaufwand wird minimiert, es 
gibt keinen redundanten Code und niemand braucht zu wissen, wie diese Dateien heißen. Vernünf-
tiges Arbeitet hat immer nur Vorteile: 

//TSOBAT EXEC PGM=IKJEFT01   Aufruf des TSO 

//SYSTSIN  DD *                               Tastatur 

  %MYREXX 

//SYSTSOUT DD SYSOUT=*                        Bildschirm 

//SYSEXEC  DD DISP=SHR,DSN=SP1.MYREXX.EXEC    Meine REXXe 

//REXXLIBS INCLUDE MEMBER=$REXXLIB 

Und so sähe das Member $REXXLIB aus (es muss lediglich in einer JES-Proclib gefunden werden): 

//         DD DISP=SHR,DSN=SYS1.ISPF.SISPEXEC 

//         DD DISP=SHR,DSN=SYS1.SBPXEXEC 

//         DD DISP=SHR,DSN=ABC.COMPANY.ALLG.EXEC 

//         DD DISP=SHR,DSN=ABC.COMPANY.TOOLS.EXEC 

//         DD DISP=SHR,DSN=ABC.COMPANY.FUNC.EXEC 

Vergleichbares kann man selbstverständlich auch mit privaten Proclibs tun. Und was sonst noch so 
benötigt wird (ISPF-Umgebung, LoadLib-Erweiterungen und dergleichen) lässt sich auf die gleiche 
Art und Weise behandeln. 

 

7.10. Ablaufsteuerung 

Es gibt zwei verschiedene Varianten, Abläufe in einer JCL zu steuern. Das heißt: wir wollen bestim-
men, welcher Step unter welchen Bedingungen laufen soll, oder eben nicht. Was es immer schon 
gab, war der COND-Parameter, welcher in der JOB- oder in der EXEC-Anweisung mit unterschied-
licher Syntax kodiert werden kann. Was relativ neu ist (erst seit 20 Jahren möglich – das musste 
einfach sein), ist die Möglichkeit, mit IF-THEN-ELSE Konstrukten zu arbeiten. Diese sind sehr viel 
einfacher zu verstehen und erlauben viel granularere Abfragen, als dies in COND möglich ist. Oft 
wäre der Aufwand gerechtfertigt, die COND-Bedingungen in alter JCL durch die "neuere" Methode 
zu ersetzen. .Aber wer wagt sich schon an bestehende Abläufe? 

Zur Veranschaulichung der beiden Varianten gehen wir von folgender Aufgabenstellung aus: 

1. Step2 läuft nur, wenn Step1 keinen RC >= 8 liefert 
2. Step3 läuft, wenn Step1 einen RC 0 und Step2 einen RC <= 4 liefert 
3. Step4 darf nicht laufen, wenn Step3 einen RC > 4 hat 
4. Step5 läuft nur bei vorausgegangenem ABEND 

 

7.10.1. COND 

Das Erste ekelige an COND ist, dass hier keine Laufbedingung, sondern eine Abbruchbedingung 
formuliert werden muss. Dies ist konträr zur "normalen" Denkweise "normaler" Menschen. Das 
zweite ist, dass bei dem zu formulierenden Ausdruck (TRUE = Abbruch, FALSE = Verarbeitung) ein 
Vergleichswert kodiert wird, welcher gegen den Return Code der Steps verglichen wird. Die meisten 
"normalen" Menschen unterstellen aber, dass es sich beim Vergleichswert um den RC handelt. Das 
ist falsch. In der Praxis hat sich bewährt, eine Condbedingung ohne viel darüber nachzudenken, 
hinzuschreiben und sie danach zu negieren. Klingt blöd, ist aber so. 

Und das wäre die Lösung: 

//STEP1 EXEC PGM=? 

//STEP2 EXEC PGM=?,COND=(8,LE,STEP1) 

//STEP3 EXEC PGM=?,COND=((0,NE,STEP1),(4,LT,STEP2)) 

//STEP4 EXEC PGM=?,COND=(4,LT,STEP3) 

//STEP5 EXEC PGM=?,COND=ONLY 

7.10.2. IF-THEN-ELSE 

Das IF-THEN-ELSE Konstrukt bietet gegenüber der Condanweisung nur Vorteile.  

▪ Formuliert wird eine Laufbedingung, was der normalen Denkweise deutlich besser entge-
genkommt 

▪ Für eine Abfrage kann ein NEIN-Zweig kodiert werden 
▪ Die möglichen Prüfungen sind sehr viel granularer. 
▪ Vieles kann durch den Condparameter gar nicht geprüft werden 

Mit IF kann nicht nur auf den RC eines oder mehrerer Steps eingegangen werden. Es können auch 
ABEND-Codes abgefragt werden. Zudem lässt sich auch feststellen, ob ein Step überhaupt verar-
beitet wurde (könnte ja aufgrund einer Abbruchbedingung übersprungen worden sein (Details hierzu 
sind im Referenzteil (Band 2) beschrieben). 

Und so sähe die Lösung aus: 

//STEP1    EXEC PGM=? 

// IF RC < 8 THEN 

//STEP2 EXEC PGM=? 

// ENDIF 

// IF STEP1.RC = 0 AND STEP2.RC <= 4 THEN 

//STEP3 EXEC PGM=? 

// ENDIF 

// IF STEP3.RC <= 4 THEN 

//STEP4 EXEC PGM=? 

// ENDIF 

// IF ABEND THEN 

//STEP5 EXEC PGM=? 

// ELSE 

//STEP6    EXEC PGM=? 

// ENDIF 

7.11. JCL-Prozeduren 

Eine JCL-Prozedur ist eine Folge von JCL-Anweisungen, die in dieser oder ähnlicher Art häufig 
benutzt werden soll. Diese Anweisungen können aus mehreren Programmaufrufen, so genannten 
Prozedursteps, bestehen und werden als Gesamtes durch Angabe des Prozedurnamens (Member-
name der Prozedur in einer JCL-Prozedurbibliothek) in der EXEC-Anweisung in einen JOBSTREAM 
eingeflochten. 

Das Zusammenführen kodierter JCL und aufgerufenen JCL-Prozeduren erfolgt während der Con-
version-Phase des JES. Die Anweisungen aus der Prozedur werden unverändert, oder modifiziert, 
übernommen. Ebenso können JCL-Anweisungen an eine bestehende Prozedur angehängt, oder 
existente annulliert werden. 

Die Veränderung von JCL-Prozeduren bezeichnet man als Expansion. Das Resultat des Zusam-
menführens und eventuellen Veränderns von Prozedurteilen ist der ablauffähige JOB im z/OS. 

Der größte Vorteil in der Anwendung einer JCL-Prozedur liegt in der Verwendung von symbolischen 
Variablen, deren Inhalte mit Standardwerten vorbesetzt werden können, oder die bei Aufruf der Pro-
zedur in der JCL bekanntgegeben werden. 

Um die Herkunft von JCL-Anweisungen im JOBLOG-Protokoll erkennen zu können (JOB oder PRO-
ZEDUR) führt das Spoolsystem in den Spalten 1 bis 3 folgende Kennzeichnung durch: 

Stelle 1-3 Bedeutung 

XX Unveränderte JCL aus einer katalogisierten Prozedur 

X/ Modifizierte JCL aus einer katalogisierten Prozedur 

XX* Kommentaranweisung aus einer katalogisierten Prozedur 

++ Unveränderte JCL aus einer Instream-Prozedur 

+/ Modifizierte JCL aus einer Instream-Prozedur 

++* Kommentaranweisung aus einer Instream-Prozedur 

*** Kommentare oder JES-Anweisungen 

7.11.1. JES-Proclib-Verkettung 

Innerhalb des Spoolsystems werden Verkettungen von Proclib-Bibliotheken unter verschiedenen 
DD-Namen im JES durchgeführt. 

DD-Name Zuständig für 

PROC00 Cataloged Procedures 
JCL welche über die Anweisung //name EXEC PROC=procname in einer JCL ange-
zogen wird 

PROC01 Started Tasks 
Werden über das MVS-Kommando START gestartet. Für gewöhnlich geschieht das 
bereits bei IPL über die Systemautomation. Im Allgemeinen sind dies Verarbeitungen 
wie beispielsweise DB2 – eine Art von sog. Listener (ein Programm, welches gestar-
tet wird und so lange aktiv bleibt, bis es kontrolliert beendet wird). 

PROC02 LOGON-Prozeduren 
JCL, welche die Grundlage für das LOGON darstellt. Ruft im Allgemeinen das 
IKJEFTxx auf und übergibt als Parameter den Namen eines REXX-Programms (the-
oretisch auch einer CLISTe), welches den weiteren Verlauf der Anmeldung steuert. 

 

Das Problem mit Cataloged Procedures bestand in früheren Zeiten darin, dass diese JCL-State-
ments zwingend in einer JES-Proclib liegen mussten und nur eine handvoll Mitarbeiter Schreibrechte 
an diesen Bibliotheken besaßen (aus gutem Grund). Seit geraumer Zeit können JCL-Prozeduren 
aber auch aus privaten Bibliotheken geladen werden, sofern eine Angabe über die entsprechenden 
Dateien in der JCL vorhanden ist. 

7.11.2. Beispiel Prozedur 

Wir betrachten zunächst, wie eine Prozedur aussehen kann, welche IEBGENER aufrufen soll.  

Die Eingabedatei ist als Nullstring vorbelegt, da es keinen Sinn ergibt, dafür ein Default zu definieren. 
Der Output wird per Default in die JES-Output-Queue H geschrieben. Steuerbefehle (SYSIN) und 
das Verarbeitungsprotokoll (SYSPRINT) werden auf DUMMY gesetzt. 

//GENER#1 PROC SYSOUT='H',INDSN= 

//GENER1 EXEC PGM=IEBGENER 

//SYSUT1   DD DSN=&INDSN,DISP=SHR 

//SYSUT2   DD SYSOUT=&SYSOUT 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

Der simple Aufruf mit Übergabe des Dateinamens sieht so aus: 

//MYPROC JCLLIB ORDER=(SP1.MYPROCS.CNTL) 

//DRUCK EXEC PROC=GENER1,INDSN='SP1.EDU.DATA(ADR1)' 

Der Aufruf der Prozedur mit umfangreicheren modifizierten Angaben könnte folgendermaßen aus-
sehen: 

//MYPROC JCLLIB ORDER=(SP1.MYPROCS.CNTL) 

//DRUCK EXEC PROC=GENER1,INDSN='SP1.EDU.DATA(ADR1)' 

//GENER1.SYSUT2 DD DSN=SP1.DEMOSEQ.DATA, 

// DISP=(NEW,CATLG),SPACE=(... 

//GENER1.SYSIN DD * 

  GENERATE MAXFLDS=3 

  RECORD FIELD=(feld-1-angaben), 

         FIELD=(feld-2-angaben), 

         FIELD=(feld-3-angaben) 

/* 

//SYSPRINT DD SYSOUT=* 

Der Name der IEBGENER-Eingabedatei wird als Parameter übernommen. Anstatt, wie in der Pro-
zedur vorgesehen, wird der Output (SYSUT2) nicht nach SYSOUT, sondern in eine Datei geschrie-
ben. Die Daten werden nicht 1:1 verarbeitet, weswegen Steuerbefehle über SYSIN geliefert werden 
und das Verarbeitungsprotokoll soll dorthin ausgegeben werden, wo auch das JOBLOG liegt. 

7.12. Utilities 

Es ist immer wieder erstaunlich, wie häufig Leute Programme schreiben, wenn ein kleines Problem 
auftaucht. Und warum tun sie das? Weil sie es können. 

Sehr viele Alltagsprobleme könnte man aber genauso gut mit den Dienstprogrammen lösen, welche 
so wie so im Haus sind. Was wäre denn so schlimm, wenn man sich in die wesentlichen Funktionen 
von einigen dieser Utilities einlesen würde. Das Sortprogramm der IBM (heißt derzeit DFSORT) ist 
eines der bestoptimierten und vielseitigsten Programme, die ich kennengelernt habe. Gut, ein Ma-
nual von 1.200 Seiten auswendig zu lernen ist nicht jedermanns Sache. Aber einen groben Überblick 
zu bekommen, was das Programm zu leisten vermag, das könnte jeder. Für Details, wenn es um 
die Syntax von Steuerbefehlen geht, kann dann immer noch das Handbuch zu Rate gezogen wer-
den. 

7.12.1. Die gängigsten Utilities 

Programmname Leistungsumfang (rudimentär) 

IEBGENER Simples Programm für Ein-/Ausgabe von Daten. Im Bedarfsfall können die 
Eingangsdaten manipuliert werden. Feldumbauten, selektive Auswahl von 
Feldern und vieles mehr. Kann PS-Dateien oder einzelne Member einer PO-
Datei verarbeiten. 

IEBCOPY Alle möglichen Funktionen im Umgang mit PO-Dateien. Gesamtes oder se-
lektives Kopieren von Membern. Ausschlussverfahren sind möglich. Kann 
eine PO-Datei auf Tape abbilden (oder auf ein sequentielles Dataset schrei-
ben) und auf dem Rückweg daraus wieder eine PO-Datei erzeugen. 

IDCAMS Für alles was mit VSAM zu tun hat. Kann auch GDGs erstellen. Wird für Un-
load/Reload-Verfahren benutzt und kennt eine Reihe von Befehlen, die uns 
aus dem TSO vertraut sind (Rename oder Delete auf Dateien und derglei-
chen mehr). 

ADRDSSU Kommt aus dem Werkzeugkasten von SMS (Storage Management System). 
HSM beispielsweise benutzt das Programm für viele Aktionen im Hintergrund. 
Damit lassen sich unter anderem beliebige Dateien in eine äußere Hülle pa-
cken und (evtl. an einem anderen Standort) wieder entpacken. Wird auch 
gerne benutzt, um Dateien mit anderen HLQs zu versehen. 

IKJEFTxx Das ist das TSO der IBM. Generell lässt sich damit alles, was im Foreground 
einer TSO-Session möglich ist, auch im Batch verarbeiten. Bis zu einem ge-
wissen Grad sogar Dialog-Anwendungen. 

PKZIP Kennt jedermann vom PC. Das Programm ist auch auf dem Host verfügbar 
und von den meisten größeren Unternehmen lizensiert 

DFSORT Kann neben SORT und MERGE auch für sequentielle Verarbeitung benutzt 
werden. Hier die Vorteile aufzuzeigen würde jedem Rahmen sprengen. Vom 

Umblocken von Daten über die Steuerung per Einschluss- oder Ausschluss-
bedingungen, Feldumbauten Einfügen oder Entfernen von Feldern bis hin zu 
logisch komplexen Abfragen (Wenn dies oder jenes in den Daten vorkommt 
soll folgendes passieren) ist nahezu alles möglich, was einem im Alltag so 
unterkommt. 
Was die Datenformate angeht, welche DFSORT verarbeiten kann, gibt es 
praktisch keine Einschränkungen. Egal ob PS, PO, VSAM, Spanned Re-
cords, SMF-Sätze oder was auch immer sonst. DFSORT kann es. 
Sort kann Dinge anstelle von REXX in einem Bruchteil der Zeit erledigen und 
es ist unerheblich wie viele Millionen von Records verarbeitet werden sollen. 
Mit REXX wird das irgendwann zu einem richtigen Problem. Nicht, dass es 
nicht ginge, aber es wird sehr teuer. 

ICETOOL Ist ein Teil von DSORT und schon seit weit mehr als 30 Jahren. Leider ist das 
Programm weitgehend unbekannt. Es erlaubt mit einer sehr einfachen Syntax 
eine simple Lösung auch für komplexe Problemstellungen. Mit ICETOOL kön-
nen im Wesentlichen auch alle DFSORT-Steuerbefehlt benutzt werden 

 

Und, und, und… 

Ich habe mich schon oft gefragt, warum sich bisher noch niemand die Mühe gemacht hat, zu diesem 
Thema mal ein gutes Buch zu schreiben.  

Die nachfolgenden Beispiele sind nach Aufgabenstellung geordnet, da oftmals mit unterschiedlichen 
Programmen zum gleichen Ergebnis gelangt werden kann und so die verschiedenen Ansätze besser 
nebeneinander betrachtet werden können. 

7.12.2. Umbau einer Datenstruktur. 

Eingabe: Stelle 30 Länge 51 Anschrift 

         Stelle 14 Länge 16 Nachname 

         Stelle  1 Länge 13 Vorname 

         Stelle 81 Länge 52 Restdaten 

Ausgabe: Anschrift – Nachname – Vorname – Restdaten unverändert 

7.12.2.1. IEBGENER 

//UMBAU    EXEC PGM=IEBGENER 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA(ADRTEST) 

//SYSUT2   DD SYSOUT=* 

//SYSPRINT DD SYSOUT=* 

//SYSIN    DD * 

  GENERATE MAXFLDS=4 

  RECORD FIELD=(51,30,,1), 

         FIELD=(16,14,,52), 

         FIELD=(13,1,,68), 

         FIELD=(52,81,,81) 

7.12.2.2. SORT 

//UMBAU    EXEC PGM=SORT 

//SORTIN   DD DISP=SHR,DSN=SP1.EDU.DATA(ADRTEST) 

//SORTOUT  DD SYSOUT=* 

//SYSOUT   DD SYSOUT=* 

//SYSIN    DD * 



Inside z/OS Seite 399 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

  SORT FIELDS=(COPY) 

  OUTREC FIELDS=(30,51,14,16,1,13,81,52) 

7.12.3. Reorganisieren einer PO-Datei 

Ziel ist, nach dem Entfernen einiger Member (alle, die mit MISCH beginnen) eine PO-Datei zu reor-
ganisieren. Die Extents sollen aufgelöst werden und Primary und Secondary Space "vernünftig" be-
stimmt werden. 

7.12.3.1. IEBCOPY (1) 

//REORG1   EXEC PGM=IEBCOPY 

//PO#IN    DD DSN=SP1.EDU.DATA,DISP=(OLD,DELETE) 

//PO#OUT   DD DISP=(,PASS),UNIT=VIO,SPACE=(CYL,(1,10,50)) 

//SYSUT3   DD UNIT=VIO,SPACE=(CYL,10) 

//SYSUT4   DD UNIT=VIO,SPACE=(CYL,10) 

//SYSPRINT DD SYSOUT=* 

//SYSIN    DD * 

  COPY OUTDD=PO#OUT,INDD=PO#IN 

  EXCLUDE MEMBER=(MISCH*) 

//REORG2   EXEC PGM=IEBCOPY 

//PO#IN    DD DSN=*.REORG1.PO#OUT,DISP=OLD 

//PO#OUT   DD DSN=SP1.EDU.DATA,DISP=(NEW,CATLG),UNIT=SYSDA, 

//            SPACE=(CYL,(1,10,50)) 

//SYSUT3   DD UNIT=VIO,SPACE=(CYL,10) 

//SYSUT4   DD UNIT=VIO,SPACE=(CYL,10) 

//SYSPRINT DD SYSOUT=* 

//SYSIN    DD * 

  COPY OUTDD=PO#OUT,INDD=PO#IN 

Es besteht folgendes Risiko: Endet der erste Step mit einem RC 12, wurde der Output nicht erzeugt, 
aber die Eingabe dennoch gelöscht. Merke: Ein RC > 0, egal welch schlimmes Ereignis er meldet, 
ist ein NORMALES! Ende. 

7.12.3.2. IEBCOPY (2) 

//CLEANUP  EXEC PGM=EQQDELDS 

//SYSPRINT DD SYSOUT=* 

//REORG1   EXEC PGM=IEBCOPY 

//PO#IN    DD DSN=SP1.EDU.DATA,DISP=OLD 

//PO#OUT   DD DSN=SP1.EDUNEU.DATA,DISP=(NEW,CATLG),UNIT=SYSDA, 

//            LIKE=SP1.EDU.DATA 

//SYSUT3   DD UNIT=VIO,SPACE=(CYL,10) 

//SYSUT4   DD UNIT=VIO,SPACE=(CYL,10) 

//SYSPRINT DD SYSOUT=* 

//SYSIN    DD * 

  COPY OUTDD=PO#OUT,INDD=PO#IN 

  EXCLUDE MEMBER=(MISCH*) 

//REORG1OK IF REORG1.RC = 0 THEN 

//REORG2   EXEC PGM=IDCAMS 

//SYSPRINT DD SYSOUT=* 

//SYSIN    DD * 

  DELETE SP1.EDU.DATA NONVSAM 

  ALTER SP1.EDUNEU.DATA NEWNAME(SP1.EDU.DATA) 

// ENDIF 

Hier gibt es mehr Sicherheit. Die alte Datei bleibt zunächst erhalten. Für das Kopieren wird eine 
neue Datei angelegt. Nur wenn der erste Step mit einem RC 0 endete, erfolgt im zweiten Step per 
IDCAMS das Löschen der Alten Datei und anschließend das Umbenennen der neuen Datei auf den 
Namen der alten. 

7.12.3.3. IKJEFTxx 

Warum kompliziert, wenn man es einfach haben kann? 

//REORG EXEC PGM=IKJEFT01 

//SYSTSPRT DD SYSOUT=* 

//SYSTSIN  DD * 

  DEL 'SP1.EDU.DATA(MISCH*)' 

  HBACKDS  'SP1.EDU.DATA' WAIT 

  HMIGRATE 'SP1.EDU.DATA' WAIT 

  HRECALL  'SP1.EDU.DATA' WAIT 

/* 

Hier machen wir unsere Arbeit im TSO-Batch und nutzen wir die Vorzüge von HSM.  

Der Delete-Befehl entfernt alle Member, die mit MISCH beginnen. 

Per HSM führen wir der Reihe nach, einen Backup, Migrate und Recall durch. Durch die Angabe 
des Parameters WAIT bekommen wir eine synchrone Verarbeitung, da anderenfalls ein könnte, dass 
sich die Befehle an den HSM überholen könnten (HSM arbeitet ja prinzipiell asynchron). Nach dem 
Recall ist sichergestellt, dass der Gesamtplatzbedarf im Primary Space abgebildet ist und das Da-
taset keine Extents mehr aufweist. 

7.12.4. GDGs (Generation Data Groups) 

GDGs stellen eine spezielle Form der Sicherungsfristen von Dateien dar, können aber nicht mit 
normalen DISP-Angaben erzeugt werden. Für die Anlage eines GDGs muss das Programm 
IDCAMS benutzt werden. Bei einem definierten Limit von 3 werden die jeweils letzten drei Versionen 
eines GDGs gehalten. Wir die vierte erstellt und hat dies geklappt, wird die erste verworfen. 

Angesprochen werden die jeweiligen Generationen wie folgt: 

DSN=name.des.gdgs(+0)  Die aktuelle Version 

DSN=name.des.gdgs(-1)  Die Vergängerversion 

DSN=name.des.gdgs(-2)  Die davor etc. 

DSN=name.des.gdgs(+1)  Die neu zu erstellende Version 

DSN=name.des.gdgs(+2)  Sofern ein Job mehrere Neuversionen erstellt 

Eine detailliertere Beschreibung ist in Band 2 bei der Beschreibung des DSN-Parameters oder im 
Manual zu IDCAMS zu finden. 

Die folgenden Beispiele sollen folgendes bewerkstelligen: 

1. Erstellen eines GDGs mit fünf Generationen 
2. Erstellen einer neuen Generation mit IEBGENER (kann mehrmals laufen) 
3. Schreiben einer jeweils neuen Generation in einem Job mit zwei Steps (hier gibt es abhängig 

von Parametern in der JOB-Anweisung unterschiedliche Möglichkeiten) 
4. Verarbeiten aller existierenden Generationen konkateniert in der Reihenfolge ihrer Erstellung 

7.12.4.1. GDG erstellen 

//BUILDGDG EXEC PGM=IDCAMS 

//SYSPRINT DD SYSOUT=* 

//SYSIN    DD * 

  DEFINE GDG ( - 

    NAME(SP1.EDU.GDG1DEMO) – 

    LIMIT(5) – 

    NOEMPTY – 

    ) 

7.12.4.2. Schreiben einer neuen Generation 

//GDGDEMO1 EXEC PGM=IEBGENER 

//* ************************************************************** 

//* DER NAME DES GDGS IST ABHäNGIG VON DER ANGABE IM JOB GDG1IDCA 

//* ************************************************************** 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA(ADR1)  

//SYSUT2   DD DSN=SP1.EDU.GDG1DEMO(+01), 

//            DISP=(NEW,CATLG), 

//            SPACE=(TRK,(300,90),RLSE), 

//            DCB=(MODEL,RECFM=FB,LRECL=132,BLKSIZE=0) 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

7.12.4.3. Schreiben von zwei neuen Generationen in einem JOB 

//GDGDEMO2 EXEC PGM=IEBGENER 

//* ************************************************************** 

//* DER NAME DES GDGS IST ABHäNGIG VON DER ANGABE IM JOB GDG1IDCA 

//* ************************************************************** 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA(ADR1)  

//SYSUT2   DD DSN=SP1.EDU.GDG1DEMO(+1), 

//            DISP=(NEW,CATLG), 

//            SPACE=(TRK,(300,90),RLSE), 

//            DCB=(MODEL,RECFM=FB,LRECL=132,BLKSIZE=0) 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

//GDGDEMO3 EXEC PGM=IEBGENER 

//* ************************************************************** 

//* DER NAME DES GDGS IST ABHäNGIG VON DER ANGABE IM JOB GDG1IDCA 

//* ************************************************************** 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA(ADR1)  

//SYSUT2   DD DSN=SP1.EDU.GDG1DEMO(+2), 

//            DISP=(NEW,CATLG), 

//            SPACE=(TRK,(300,90),RLSE), 

//            DCB=(MODEL,RECFM=FB,LRECL=132,BLKSIZE=0) 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

Da die Interpretation der JCL vor Verarbeitung für den Gesamtjob durchgeführt erfolgt, wird im ersten 
Step über +1 und im zweiten Step über +2 der jeweilige Name der Ausgabedatei bestimmt. Würde 
in beiden Steps +1 benutzt werden, würde sich in beiden Fällen der gleiche Dateiname hinter der 
DSN-Angabe verbergen und dies würde zum Abbruch der Verarbeitung führen. 

7.12.4.4. Schreiben von zwei neuen Generationen in einem JOB 

//SP1A     JOB (4711),Notify=&SYSUID,GDGBIAS=STEP 

//GDGDEMO4 EXEC PGM=IEBGENER 

//* ************************************************************** 

//* DER NAME DES GDGS IST ABHäNGIG VON DER ANGABE IM JOB GDG1IDCA 

//* ************************************************************** 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA(ADR1)  

//SYSUT2   DD DSN=SP1.EDU.GDG1DEMO(+1), 

//            DISP=(NEW,CATLG), 

//            SPACE=(TRK,(300,90),RLSE), 

//            DCB=(MODEL,RECFM=FB,LRECL=132,BLKSIZE=0) 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

//GDGDEMO2 EXEC PGM=IEBGENER 

//* ************************************************************** 

//* DER NAME DES GDGS IST ABHäNGIG VON DER ANGABE IM JOB GDG1IDCA 

//* ************************************************************** 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA(ADR1)  

//SYSUT2   DD DSN=SP1.EDU.GDG1DEMO(+1), 

//            DISP=(NEW,CATLG), 

//            SPACE=(TRK,(300,90),RLSE), 

//            DCB=(MODEL,RECFM=FB,LRECL=132,BLKSIZE=0) 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

Über den Parameter GDGBIAS in der JOB-Anweisung kann geregelt werden, ob die Namens-
vergabe für neue Generationen eines GDGs auf JOB- oder Step-Ebene durchgeführt werden soll. 
GDGBIAS als Parameter in der JOB-Anweisung kam mit zOS 2.1 

7.12.4.5. Verarbeitung aller Generationen eines GDGs 

Würde man die einzelnen Generationen des GDGs direkt ansprechen wollen, ergäbe sich folgende 
Angabe in der JCL: 

//ddname   DD DISP=SHR,SP1.EDU.GDGDEMO(-4) 

//         DD DISP=SHR,SP1.EDU.GDGDEMO(-3) 

//         DD DISP=SHR,SP1.EDU.GDGDEMO(-2) 

//         DD DISP=SHR,SP1.EDU.GDGDEMO(-1) 

//         DD DISP=SHR,SP1.EDU.GDGDEMO(0) 

 

Oder 

 

//ddname   DD DISP=SHR,SP1.EDU.GDGDEMO.G0001V00 

//         DD DISP=SHR,SP1.EDU.GDGDEMO.G0002V00 

//         DD DISP=SHR,SP1.EDU.GDGDEMO.G0003V00 

//         DD DISP=SHR,SP1.EDU.GDGDEMO.G0004V00 

//         DD DISP=SHR,SP1.EDU.GDGDEMO.G0005V00 

Seit zOS 2.1 gibt es dafür folgende Möglichkeit: 

//SP1A     JOB 

//GDGDEMO4 EXEC PGM=IEBGENER 

//* ************************************************************** 

//* DER NAME DES GDGS IST ABHäNGIG VON DER ANGABE IM JOB GDG1IDCA 

//* ************************************************************** 

//SYSUT1   DD DISP=SHR,DSN=SP1.EDU.DATA,GDGORDER=FIFO 

//SYSUT2   DD SYSOUT=* 

//SYSIN    DD DUMMY 

//SYSPRINT DD DUMMY 

7.12.4.6. Löschen eines GDGs 

//DELGDG   EXEC PGM=IDCAMS 

//* *************************************************************** 

//* DELETE AUF DEN GDG-KATALOGEINTRAG KANN ERST ERFOLGEN, 

//* WENN KEIN DATASET DES GDGS MEHR EXISTIERT. 

//* 

//SYSPRINT DD SYSOUT=* 

//SYSIN    DD * 

  DELETE (SP1.EDU.GDG1DEMO) GDG FORCE 

  DELETE (SP1.EDU.GDG2DEMO) GDG FORCE 

/* 

7.12.5. Dateien einpacken 

Generell stehen uns dafür mehrere Möglichkeiten zur Verfügung, von denen hier drei aufgezeigt 
werden sollen. 

  



Inside z/OS Seite 404 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

7.12.5.1. PKZIP 

PKZIP oder eines der gleichwertigen Produkte wie 7ZIP, RAR und dergleichen, sind jedem PC-
Benutzer bekannt. Viele Unternehmen offerieren PKZIP auch auf dem Mainframe. Dort gibt es im 
Übrigen auch eine ISPF-Oberfläche, deren größter Vorteil ist, sich die Inhalte einer ZIP-Datei an-
schauen zu können, ohne sie entpacken zu müssen. 

Die Parameter von PKZIP sind sehr umfangreich. Wer das Produkt im Einsatz hat, verfügt auch über 
das entsprechende Manual. Im Bedarfsfall einfach mal den Produktverantwortlichen fragen. 

Für den folgenden Job war die Anforderung, das ZIP-File anschließend auf den PC zu laden und 
dort lesen zu können. Aus diesem Grund wird das Ausgabeformat ASCII gewählt. 

//ARCHDEL  EXEC PGM=IDCAMS 

//* ******************************************************************* 

//* Löschen des alten ARCHIVE-Datasets 

//* ******************************************************************* 

//SYSPRINT DD SYSOUT=* 

//SYSIN    DD * 

  DEL SP1.EDUSAVE.ZIP 

  SET MAXCC=0 

/* 

//PKZIP    EXEC PGM=PKZIP,REGION=4096K 

//* ******************************************************************* 

//* Zippen von Dateien 

//* PKZIP läuft in der Regel autorisiert, deshalb Steplib 

//* ------------------------------------------------------------------- 

//STEPLIB  DD DISP=SHR,DSN=pkzip_apf_lib Sofern nötig 

//         DD DISP=SHR,DSN=library_mit_translation_tables 

//* ------------------------------------------------------------------- 

//SYSPRINT DD SYSOUT=* 

//SYSIN    DD * 

-ECHO 

-TRAN(CULASCII) 

-METHOD(SUPERFAST) 

-ARCHBLKSIZ(27795) 

-ARCHUNIT(SYSALLDA) 

-ARCHTYPE(FB) 

-ARCHSPACE(TRK) 

-ARCHPRIMARY(30) 

-ARCHSECONDARY(15) 

-TRANSLATE_TABLE_DATA(TRTSTD1) 

-TRANSLATE_TABLE_FILEINFO(TRTSTD1) 

-ARCHIVE(SP1.EDUSAVE.ZIP) 

SP1.EDU.EXEC(*) 

SP1.EDU.DATA(*) 

SP1.EDU.PANELS(*) 

/* 

7.12.5.2. ADRDSSU Einpacken 

//CLEANUP  EXEC PGM=EQQDELDS 

//SYSPRINT DD SYSOUT=* 

//MYDSSU   EXEC PGM=ADRDSSU 

//SYSPRINT DD SYSOUT=* 

//DUMPOUT  DD DSN=SP1.DSSUDUMP.DATA, 

//            DISP=(NEW,CATLG,DELETE), 

//            DCB=BLKSIZE=32000, 

//            SPACE=(CYL,(100,100),RLSE) 

//DUPL     DD DUMMY,DCB=(BLKSIZE=32760),DISP=SHR 

//SYSIN    DD * 

  DUM – 

      OUTDDNAME(DUMPOUT)     - 

      TOLERATE(ENQF)         - 

      DS(INCLUDE(            - 

         SP1.EDU.COBOL       - 

         SP1.EDU.DATA        - 

         SP1.EDU.EXEC        - 

         SP1.EDU.PANELS      - 

        )) 

/*  

7.12.5.3. ADRDSSU Auspacken 

Das Entpacken von gedumpten Dateien erlaubt gegebenenfalls auch den Wechsel von HLQs, wie 
das folgende Beispiel zeigt. Auf diese Weise können beim Entpacken Dateinamen erzeugt werden 
auf welche der ursprüngliche Eigentümer aufgrund der RACF-Definitionen nicht mehr zugreifen darf. 
Das folgende Beispiel zeigt das Umbenennen beim Auspacken, bei dem sich der zweite Qualifier 
der Dateien ändert: 

//UNPACK   EXEC PGM=ADRDSSU 

//SYSPRINT DD SYSOUT=* 

//DSSUIN   DD DISP=OLD,DSN=SP1.DSSUDUMP.DATA 

//DSSUOUT  DD SPACE=(CYL,(100,100)) 

//SYSIN    DD * 

  RESTORE INDD(DSSUIN) OUTDD(DSSUOUT)     - 

      DS(INCLUDE(**))                     - 

      RENUNC(SP1.EDU.**, SP1.DSSUOUT.**)  - 

      NSC                                 - 

      NMC                                 - 

      CATALOG 

/* 

7.12.5.4. XMIT 

Ist zwar eines der privilegierten TSO-Commands, wird aber ebenfalls sehr gerne benutzt, um Daten 
zwischen Mainframern auszutauschen. Das spezielle Format (XMIT) ist grundsätzlich nur auf dem 
Host gebräuchlich, es gibt mittlerweile im Internet frei verfügbar sogenannte XMIT-Manager, mit de-
nen diese Daten (EBCDIC) auch auf dem PC gelesen werden können. 

Der Umgang damit ist sehr "individuell". Ich selbst benutze das gerne für folgendes: 

Je ein Set von REXXen, Panels, Skeletons (und so weiter) wird im XMIT-Format in jeweils einen 
PO-Member geschrieben. Damit das Ganze vernünftig transportabel ist, wird das Ergebnis per XMIT 
in eine sequentielle Datei kopiert. Diese kann per Download auf den PC geholt und gegebenenfalls 
per Mail verschickt werden (das Herunterladen sollte binär erfolgen). 

Hierzu die JCL für das 2-stufige Einpacken 

//SP1A    JOB (RZRZ),MSGCLASS=H,NOTIFY=&SYSUID 

//* ****************************************************** 

//* Erstellen XMIT-PO-Dataset 

//* ****************************************************** 

//CREARCH  EXEC PGM=IEFBR14 

//XMITPO   DD DSN=SP1.XMIT.DATA, 

//            DISP=(NEW,CATLG,DELETE), 

//            SPACE=(CYL,(1,5,1)),DSNTYPE=LIBRARY, 

//            DCB=(LRECL=80,RECFM=FB,BLKSIZE=0) 

//* ****************************************************** 

//* TSO-Aufruf für XMIT 

//* ****************************************************** 

//TSO      EXEC PGM=IKJEFT01 

//SYSTSPRT DD SYSOUT=* 

//OUT1     DD DISP=SHR,DSN=SP1.XMIT.DATA(TRYREXX) 

//OUT2     DD DISP=SHR,DSN=SP1.XMIT.DATA(TRYPAN) 

//OUT3     DD DISP=SHR,DSN=SP1.XMIT.DATA(TRYSKEL) 

//OUT4     DD DISP=SHR,DSN=SP1.XMIT.DATA(TRYMSGS) 

//OUT5     DD DISP=SHR,DSN=SP1.XMIT.DATA(TRYTABL) 

//OUT6     DD DISP=SHR,DSN=SP1.XMIT.DATA(SAMPREXX) 

//OUT7     DD DISP=SHR,DSN=SP1.XMIT.DATA(SAMPPAN) 

//OUT8     DD DISP=SHR,DSN=SP1.XMIT.DATA(SAMPSKEL) 

//OUT9     DD DISP=SHR,DSN=SP1.XMIT.DATA(SAMPCNTL) 

//OUT10    DD DISP=SHR,DSN=SP1.XMIT.DATA(FUNREXX) 

//OUT11    DD DISP=SHR,DSN=SP1.XMIT.DATA(FUNPAN) 

//SYSTSIN  DD * 

  XMIT NODE01.LOCAL OUTDD(OUT1)  DA('SP1.TRYOUT.EXEC') 

  XMIT NODE01.LOCAL OUTDD(OUT2)  DA('SP1.TRYOUT.PANELS') 

  XMIT NODE01.LOCAL OUTDD(OUT3)  DA('SP1.TRYOUT.SKELS') 

  XMIT NODE01.LOCAL OUTDD(OUT4)  DA('SP1.TRYOUT.MSGS') 

  XMIT NODE01.LOCAL OUTDD(OUT5)  DA('SP1.TRYOUT.TABLES') 

  XMIT NODE01.LOCAL OUTDD(OUT6)  DA('SP1.SAMPLES.EXEC') 

  XMIT NODE01.LOCAL OUTDD(OUT7)  DA('SP1.SAMPLES.PANELS') 

  XMIT NODE01.LOCAL OUTDD(OUT8)  DA('SP1.SAMPLES.SKELS') 

  XMIT NODE01.LOCAL OUTDD(OUT9)  DA('SP1.SAMPLES.CNTL') 

  XMIT NODE01.LOCAL OUTDD(OUT10) DA('SP1.GANDMORE.EXEC') 

  XMIT NODE01.LOCAL OUTDD(OUT11) DA('SP1.GANDMORE.PANELS') 

  XMIT NODE01.LOCAL + 

       DATASET('SP1.XMIT.DATA') OUTDS('SP1.XMITALL.DATA') 

/* 

Die auf diese Weise erzeugte PS-Datei (SP1.XMITALL.DATA), ließe sich nun BINARY auf den PC 
laden und verschicken. Sobald das Dataset wieder BIRARY auf einen Host gestellt wurde, kann sie 
dort in 2 Schritten wieder auf ihre ursprünglichen Dateien entpackt werden. Das kann generell im 
TSO-Foreground erfolgen. Wie der erste Schritt per JCL aussehen würde, zeigt das folgende Bei-
spiel: 

//******************************************************************** 

//* RECEIVE ALL XMITTED DATASETS TO MEMBER IN A PDS 

//******************************************************************** 

//RECEIVE1 EXEC PGM=IKJEFT01 

//SYSTSPRT DD  SYSOUT=* 

//INDD1    DD  DISP=SHR,DSN=YOUR_INPUT_PS_DATASET 

//SYSTSIN  DD  * 

  RECEIVE INDD(INDD1) 

  DA('YOUR_OUTPUT_PO_DATASET') 

//******************************************************************** 

//* RECEIVE ALL MEMBERS TO ORIGINAL DATASETS 

//******************************************************************** 

//RECEIVE2 EXEC PGM=IKJEFT01 

//SYSTSPRT DD  SYSOUT=* 

//* XMITTED INPUT FILES 

//SYSTSIN  DD * 

  ??? Hier kämen die Angaben für das Entpacken der einzelnen Members 

      des im oberen Step erzeugten PO-Datasets auf die jeweilige 

      PO-Datei, welche die Member enthalten soll. Viel Spaß     

7.12.6. SORT & ICETOOL 

DFSORT und ICETOOL können auch gut kombiniert arbeiten. Soll ein ICETOOL-Step Sortbefehle 
verarbeiten, können diese unter dem DD-Namen xxxxCNTL übergeben werden. Wie dieses State-
ment genau heißt, kann beeinflusst werden. Der Default Name ist SORTCNTL, welcher im weiteren 
Verlauf benutzt wird. 

7.12.6.1. Einfügen Titelzeile und Spaltenüberschriften 

//ICETOOL2 EXEC PGM=ICETOOL 

//TOOLMSG  DD SYSOUT=* 

//DFSMSG   DD SYSOUT=* 

//EINGABE  DD DISP=SHR,DSN=SP1.EDU.DATA(ADR1) 

//AUSGABE  DD SYSOUT=* 

//TOOLIN   DD * 

  DISPLAY FROM(EINGABE) LIST(AUSGABE) – 

     TITLE('Adressverzeichnis') PAGE DATENS(DM4) TIME – 

     HEADER('Vorname')      ON(01,13,CH) – 

     HEADER('Nachname')     ON(14,16,CH) – 

     HEADER('  PLZ Ort')    ON(30,22,CH) – 

     HEADER('Straße')       ON(52,27,CH) 

7.12.6.2. Zähle Datensätze mit PLZ kleiner 2000 

//COUNT1   EXEC PGM=ICETOOL 

//* *************************************************************** 

//* Wie viele Sätze haben eine PLZ kleiner als 20000 

//* *************************************************************** 

//TOOLMSG  DD SYSOUT=* 

//DFSMSG   DD SYSOUT=* 

//EINGABE  DD DISP=SHR,DSN=SP1.EDU.DATA(ADRSORT) 

//TOOLIN   DD * 

  COUNT FROM(EINGABE) 

//SORTCNTL DD * 

  INCLUDE COND=(30,5,LT,C'20000'),FORMAT=CH 

7.12.6.3. Zähle Datensätze mit PLZ zwischen 10000 und 19999 

//COUNT2   EXEC PGM=ICETOOL 

//* *************************************************************** 

//* Wie viele Sätze haben eine PLZ zwischen 10000 und 19999 

//* *************************************************************** 

//TOOLMSG  DD SYSOUT=* 

//DFSMSG   DD SYSOUT=* 

//EINGABE  DD DISP=SHR,DSN=SP1.EDU.DATA(ADRSORT) 

//TOOLIN   DD * 

  COUNT FROM(EINGABE) 

//SORTCNTL DD * 

  INCLUDE COND=(30,5,GE,C'10000',AND,30,5,LT,C'20000'),FORMAT=CH 

7.12.6.4. Schreibe Eingabe auf mehrere Ausgaben 

//DSPRE    SET DSPRE='SP1.ICEOUT'       Es ergeben sich die Dateien: 

//DSSUF    SET DSSUF='.DATA'            SP1.ICEOUT1|2|3.DATA 

//SPACE    SET SPACE='CYL,(1,1)' 

//DISP     SET DISP=',CATLG,DELETE' 

//CLEANUP  EXEC PGM=EQQDELDS 

//SYSPRINT DD SYSOUT=* 

//MULTCOPY EXEC PGM=ICETOOL 

//TOOLMSG  DD SYSOUT=* 

//DFSMSG   DD SYSOUT=* 

//ICE#IN   DD DISP=SHR,DSN=SP1.EDU.DATA(ADRTEST) 

//ICE#OUT1 DD DISP=(&DISP),DSN=&DSPRE.1&DSSUF, 

//            SPACE=(&SPACE) 

//ICE#OUT2 DD DISP=(&DISP),DSN=&DSPRE.2&DSSUF, 

//            SPACE=(&SPACE) 

//ICE#OUT3 DD DISP=(&DISP),DSN=&DSPRE.3&DSSUF, 

//            SPACE=(&SPACE) 

//TOOLIN   DD * 

  COPY FROM(ICE#IN) TO(ICE#OUT1,ICE#OUT2,ICE#OUT3) 

7.12.6.5. Überschriften, Summenbildung, Min/Max/Durchschnitts-Werte 

//ICELIST  EXEC PGM=ICETOOL 

//TOOLMSG  DD SYSOUT=* 

//DFSMSG   DD SYSOUT=* 

//PREISIN  DD DISP=SHR,DSN=SP1.EDU.DATA(ITPREIS) 

//PREISOUT DD SYSOUT=* 

//TOOLIN   DD * 

   DISPLAY FROM(PREISIN) LIST(PREISOUT)    - 

      TITLE('Artikel- und Preisliste')     - 

      HEADER('ArtNr') ON(1,6,CH)           - 

      HEADER('Bezeichnung') ON(8,20,CH)    - 

      HEADER('Preis') ON(58,7,ZD,C2,T' €') – 

      AVERAGE('Durchschnitt')              - 

      MAXIMUM('Max-Wert')                  - 

      MINIMUM('Min-Wert')                  - 

      TOTAL('Summe') 

//SORTCNTL DD * 

   INCLUDE COND=(58,7,FS,EQ,NUM) 

7.12.6.6. Auswertung von SMF-Sätzen 

Hierzu folgendes zum besseren Verständnis: Beim Aufruf von Toolbox-Routinen werden ein SMF-
Records erzeugt. Auf diese Weise kann festgestellt werden, welche Tools in welcher Häufigkeit noch 
benutzt werden, bzw. ob es unbenutzte (Leichen) gibt. 

Diese Records werden am jeweiligen Monatsersten für den vorangegangenen Monat in eine Datei 
extrahiert. Das Beispiel zeigt eine Auswertung über einen Zeitraum von drei Monaten. 

//* **************************************** 

//* Auswertung der Applikationsaufrufe 

//* **************************************** 

//SMFHLQ   SET HLQ='SP1.SMF0000.S0002000.M' 

//MONAT1   SET M1='202110.DATA'  Oktober 2021 

//MONAT2   SET M2='202109.DATA'  September 2021 

//MONAT3   SET M3='202108.DATA'  August 2021 

//APPCOUNT EXEC PGM=ICETOOL 

//SORTCNTL DD * 

  INCLUDE COND=(33,8,EQ,C'00002100'),FORMAT=CH 

/* 

//TOOLMSG  DD SYSOUT=* 

//DFSMSG   DD SYSOUT=* 

//APPLCNT  DD SYSOUT=* 

//LOGFILE  DD DISP=SHR,DSN=&HLQ&M1 

//         DD DISP=SHR,DSN=&HLQ&M2 

//         DD DISP=SHR,DSN=&HLQ&M3, 

//SORTWK01 DD STORCLAS=ALTEMP,SPACE=(CYL,10,,CONTIG) 

//SORTWK02 DD STORCLAS=ALTEMP,SPACE=(CYL,10,,CONTIG) 

//SORTWK03 DD STORCLAS=ALTEMP,SPACE=(CYL,10,,CONTIG) 

//SORTWK04 DD STORCLAS=ALTEMP,SPACE=(CYL,10,,CONTIG) 

//TOOLIN   DD * 

  OCCUR FROM(LOGFILE) – 

        LIST(APPLCNT) – 

  HEADER('APPLICATION') ON(142,8,CH) – 

  HEADER('USAGES')      ON(VALCNT) BLANK 

/* 

Und so würde das Ergebnis aussehen: 

APPLICATION            USAGES 

-----------   --------------- 

ADB2                     5332 

APA                        98 

APQ                     78803 

APPC                        6 

AREST                      83 

BERICHT                  2389 

7.12.6.7. Geburtstagsliste (wo gibt es heute was zu feiern?) 

Die Liste zeigt die Geburtstage für den Monat Januar (vom 1. Bis zum 31.) 

//GEBLIST  EXEC PGM=SORT 

//SORTWK01 DD UNIT=WORK1,SPACE=(CYL,10) 

//SORTWK02 DD UNIT=WORK1,SPACE=(CYL,10) 

//SORTIN   DD DISP=SHR,DSN=SP1.EDU.DATA(ADRSORT) 

//SORTOUT  DD SYSOUT=* 

//SYSOUT   DD SYSOUT=* 

//SYSIN    DD * 

  SORT FIELDS=(84,2,ZD,A,81,2,ZD,A,87,4,ZD,D,14,5,CH,A) 

  INCLUDE COND=(84,2,EQ,C'01'),FORMAT=CH 

  OUTREC FIELDS=(81,11,14,16,1,13) 

7.12.6.8. Einfügen einer neuen Spalte 

Da die DD-Namen AUFGABE und HINWEIS vom Sort nicht berücksichtigt werden, kann auf diese 
Weise in der JCL dokumentiert werden, ohne jede Zeile als Kommentar ausweisen zu müssen. 

//INSERT   EXEC PGM=SORT 

//AUFGABE  DD * 

Fügen Sie vor den Feldern Nachname-PLZ-Straße jeweils 5 Byte ein. 

 

Vorname------Nachname--------PLZ---Ort-------------Straße-------------- 

----+----1----+----2----+----3----+----4----+----5----+----6----+----7- 

Adebar       Hunziker        07041 Schlangenhausen An der Feder 287 

/* 

//HINWEIS  DD * 

Unter dem DD-Namen SYMNAMES können Konstanten oder Felder einer 

Variable zugewiesen werden, welche im weiteren Verlauf unter //SYSIN 

verwendet werden können. 

 

Wollte man statt fünf nur 3 Byte einfügen, müsste nur an einer Stelle 

geändert werden. 

/* 

//SYMNAMES DD * 

#Blanks,C'     ' * Das sind 5 Byte 

/* 

//SYMNOUT  DD SYSOUT=*     Protokolliert die SYSNAMES-Definitionen 

//SYSOUT   DD SYSOUT=* 

//SORTOUT  DD SYSOUT=* 

//SORTIN   DD DISP=SHR,DSN=SP1.EDU.DATA(SORTIN#1) 

//SYSIN    DD * 

********************************************************************** 

* Alle Inputzeilen mit "*" an erster Stelle und die Überschritszeile 

* ignorieren. 

********************************************************************** 

  OMIT COND=(1,1,CH,EQ,C'*',OR,30,5,FS,NE,NUM) 

  OPTION COPY 

* SORT FIELDS=(COPY) wäre auch möglich 

  OUTREC  BUILD=(01,13,   * Vorname 

         #Blanks,14,16,   * fünf Leerzeichen und Nachname 

         #Blanks,30,6,    * fünf Leerzeichen und PLZ 

                 36,16,   * Ort 

         #Blanks,52,29,   * fünf Leerzeichen und Straße 

                 81,10)   * Geburtsdatum 

/* 

7.12.6.9. Zusammenführen von Dateien 

//SCHRITT1 EXEC PGM=SORT 

//AUFGABE  DD * 

Es gibt 3 Members, welche zusammengemischt werden sollen. 

SORTIN#2 enthält eine UserID und die dazugehörende Abteilung 

SORTIN#3 enthält eine UserID und den dazugehörenden Vor- und Nachnamen 

SORTIN#4 enthält eine UserID und die dazugehörende Telefonnummer 

 

Aus diesen drei Eingaben sollen Ausgabesätze gebildet werden, in denen 

UserID - Abteilung - Vor- und Nachname und die Telefonnummer steht. 

/* 

//HINWEIS  DD * 

Mit JOINKEYS kann man nur zwei Datasets zusammenführen, deshalb wird 

diese Aufgabe in zwei Schritten gelöst. 

(siehe DFSORT Application Programming Guide z/OS 2.4 Seite 437) 

Per REFORMAT-Anweisung kann man aus beiden Files Informationen 

in einem Ausgabesatz vereinen. (siehe Seite 447 im SORT-Manual) 

//SYSOUT   DD SYSOUT=* 

//SYMNAMES DD * 

F1_UID,1,7,CH  * Position der Userid im File 1 

F1_ABT,9,6,CH  * Position der Abteilungsbezeichnung im File 1 

F2_UID,1,7,CH  * Position der Userid im File 2 

F2_NVn,9,20,CH * Position der Vor- und Nachnamen im File 2 

/* 

//SYMNOUT  DD SYSOUT=*     Protokolliert die SYSNAMES-Definitionen 

//SORTOUT  DD DISP=(,PASS) ,SPACE=(CYL,(1,100)) wenn nötig 

//ABTEIL   DD DISP=SHR,DSN=SP1.EDU.DATA(SORTIN#2) 

//NAME     DD DISP=SHR,DSN=SP1.EDU.DATA(SORTIN#3) 

//SYSIN    DD * 

********************************************************************** 

* Zusammenführen der Datensätze über das UserID-Feld in File 1 und 2 

********************************************************************** 

  JOINKEYS F1=ABTEIL,FIELDS=(F1_UID,A) 

  JOINKEYS F2=NAME,FIELDS=(F2_UID,A) 

********************************************************************** 

* Mit REFORMAT Ausgabesatz mit UserID und Abteilung aus File 1 und 

* Vor- und Nachname aus File 2 bilden. 

********************************************************************** 

  REFORMAT FIELDS=(F1:F1_UID,F1_ABT,F2:F2_NVn) 

  OPTION COPY 

/* 

//SCHRITT2 EXEC PGM=SORT 

//SYSOUT   DD SYSOUT=* 

//SYMNAMES DD * 

F1_UID,1,7,CH  * Position der Userid im File 1 

F1_All,1,33,CH * Alle Daten aus File 1 (Output aus Step FIRST) 

F2_UID,1,7,CH  * Position der Userid im File 2 

F2_Tel,9,16,CH * Position der Telefonnummer im FIle 2 

/* 

//SYMNOUT  DD SYSOUT=*     Protokolliert die SYSNAMES-Definitionen 

//SORTOUT  DD SYSOUT=* 

//ALL      DD DISP=OLD,DSN=*.SCHRITT1.SORTOUT 

//TEL      DD DISP=SHR,DSN=SP1.EDU.DATA(SORTIN#4) 

//SYSIN    DD * 

********************************************************************** 

* Zusammenführen der Datensätze über das UserID-Feld in File 1 und 2 

********************************************************************** 

  JOINKEYS F1=ALL,FIELDS=(F1_UID,A) 

  JOINKEYS F2=TEL,FIELDS=(F2_UID,A) 

********************************************************************** 

* Mit REFORMAT Ausgabesatz mit allen Informationen aus File 1 und 

* Telefonnummer aus File 2 bilden. 

********************************************************************** 

  REFORMAT FIELDS=(F1:F1_All,F2:F2_Tel) 

  OPTION COPY 

********************************************************************** 

* Trennzeichen einfügen. 

********************************************************************** 



Inside z/OS Seite 412 
Das große TSO-REXXikon 

  
 
 

 

Kochbuch – Praxiswissen einfach erklärt 

  OUTREC BUILD=(1,7,C' - ',8,6,C' - ',14,20,C' - ',34,16) 

/* 

7.12.6.10. Erstellen einer CSV-Datei (1) 

//BUILDCSV EXEC PGM=SORT 

//AUFGABE  DD * 

Bilden Sie eine CSV-Datei, bei der die Felder 

Vorname-Nachname-PLZ-Ort-Straße-Geburtsdatum 

durch Semikolon voneinander getrennt sind. 

* 

Vorname------Nachname--------PLZ-- Ort------------ Straße-------------- 

*---+----1----+----2----+----3----+----4----+----5----+----6----+----7- 

Adebar       Hunziker        07041 Schlangenhausen An der Feder 287 

Don Hans     von Burger      07066 Karlskühe       Glocken Weg 23 A 

 

Würde jedes Feld aus nur einem Wort bestehen, würde folgendes reichen: 

   OUTREC  BUILD=(1,90,SQZ=(SHIFT=LEFT,MID=C';')) 

/* 

//HINWEIS  DD * 

Der erste Satz der Eingabedatei enthält ein paar nicht darstellbare 

Zeichen ("20"x und "30"x) welche daazu führen, dass der Output nicht 

exakt dem gewollten entspricht. Wollte man dies beheben (und gleich- 

zeitig alle anderen ungültigen Zeichen entfernen, könnte man dies über 

eine eigenen Translation-Table erreichen (siehe nächstes Beispiel) 

/* 

//SORTOUT  DD SYSOUT=* 

//SYSOUT   DD SYSOUT=* 

//SORTIN   DD DISP=SHR,DSN=SP1.EDU.DATA(SORTIN#5) 

//SYSIN    DD * 

********************************************************************** 

* Alle Eingabesätze mit "*" an Stelle 1 und die Überschriftszeile 

* ignorieren. 

**********************************************************************        

 OMIT COND=(1,1,CH,EQ,C'*',OR,30,5,FS,NE,NUM) 

 OPTION COPY 

*********************************************************************** 

* *---+----1----+----2----+----3----+----4----+----5----+----6----+---- 

* Adebar       Hunziker        07041 Schlangenhausen An der Feder 287 

* Don Hans     von Burger      07066 Karlskühe       Glocken Weg 23 A 

*********************************************************************** 

 INREC  BUILD=(1,13,SQZ=(SHIFT=LEFT,MID=C'~'), 

               14,16,SQZ=(SHIFT=LEFT,MID=C'~'), 

               30,6, 

               36,16,SQZ=(SHIFT=LEFT,MID=C'~'), 

               52,29,SQZ=(SHIFT=LEFT,MID=C'~'), 

               81,10) 

*********************************************************************** 

* Aufbereitung im INREC - Statement: 

* - Felder werden linksbündig gestaucht. 

* - Bei mehreren Wörtern werden ALLE Blanks entfernt. 

* - Mit "MID" wird eine Zeichenkette angegeben die zwischen den 

*   Wörtern eingesetzt wird. 

* *---+----1----+----2----+----3----+----4----+----5----+----6----+---- 

* Adebar       Hunziker        07041 Schlangenhausen An~der~Feder~287 

* Don~Hans     von~Burger      07066 Karlskühe       Glocken~Weg~23~A 

*********************************************************************** 

   OUTREC BUILD=(1,90,SQZ=(SHIFT=LEFT,MID=C';')) 

*********************************************************************** 

* Aufbereitung im OUTREC - Statement: 

* - Die gesamte Zeile wird linksbündig gestaucht mit ";" als 

*   Worttrennung 

* *---+----1----+----2----+----3----+----4----+----5----+----6----+---- 

* Adebar;Hunziker;07041;Schlangenhausen;An~der~Feder~287;05.10.1953 

* Don~Hans;von~Burger;07066;Karlskühe;Glocken~Weg~23~A;24.02.1983 

*********************************************************************** 

   OUTFIL FINDREP=(IN=C'~',OUT=C' ') 

*********************************************************************** 

* Aufbereitung im OUTFIL - Statement: 

* - Alle "~" werden durch Blank ersetzt 

* *---+----1----+----2----+----3----+----4----+----5----+----6----+---- 

* Adebar;Hunziker;07041;Schlangenhausen;An der Feder 287;05.10.1953 

* Don Hans;von Burger;07066;Karlskühe;Glocken Weg 23 A;24.02.1983 

*********************************************************************** 

/* 

7.12.6.11. Erstellen CSV-Datei (2 – mit Translation Table) 

//CONVERT  EXEC PGM=SORT 

//AUFGABE  DD * 

Bilden Sie eine CSV-Datei, bei der die Felder 

Vorname-Nachname-PLZ-Ort-Straße-Geburtsdatum 

durch Semikolon voneinander getrennt sind. 

* 

Vorname------Nachname--------PLZ-- Ort------------ Straße-------------- 

*---+----1----+----2----+----3----+----4----+----5----+----6----+----7- 

Adebar       Hunziker        07041 Schlangenhausen An der Feder 287 

Don Hans     von Burger      07066 Karlskühe       Glocken Weg 23 A 

/* 

//HINWEIS  DD * 

Mit ALTSEQ wird eine Translation-Table definiert. 

 

     CODE=(aabb) aa = Byte in der Eingabe soll ersetzt werden durch 

                 bb = Byte in der Ausgabe (in HEX) 

 

(siehe DFSORT Application Programming Guide z/OS 2.4 Seite 82) 

 

Mit SQZ (squeeze=quetschen) werden alle SPACES entfernt (Manual S. 138) 

/* 

//SORTIN   DD DISP=SHR,DSN=SP1.EDU.DATA(SORTIN#5) 

//SORTOUT  DD DISP=(,PASS) ,SPACE=(CYL,(1,100)) wenn nötig 

//SYSOUT   DD SYSOUT=* 

//SYSIN    DD * 

 OPTION COPY 

 INREC  BUILD=(1,90,TRAN=ALTSEQ) 

 ALTSEQ  CODE=(0040,0140,0240,0340,0440,0540,0640,0740, 

               0840,0940,0A40,0B40,0C40,0D40,0E40,0F40, 

               1040,1140,1240,1340,1440,1540,1640,1740, 

               1840,1940,1A40,1B40,1C40,1D40,1E40,1F40, 

               2040,2140,2240,2340,2440,2540,2640,2740, 

               2840,2940,2A40,2B40,2C40,2D40,2E40,2F40, 

               3040,3140,3240,3340,3440,3540,3640,3740, 

               3840,3940,3A40,3B40,3C40,3D40,3E40,3F40, 

               4040,4140,4240,4340,4440,4540,4640,4740, 

               4840,4940,4C40,4D40,4E40,4F40,5040, 

               5140,5240,5340,5440,5540,5640,5740,5840, 

               5940,5B40,5C40,5D40,5E40,5F40,6140,6240, 

               6340,6440,6540,6640,6740,6840,6940,6B40, 

               6C40,6D40,6E40,6F40,7040,7140,7240,7340, 

               7440,7540,7640,7740,7840,7940,7A40,7B40, 

               7C40,7E40,7F40,8040,8A40,8B40,8C40,8D40, 

               8E40,8F40,9040,9A40,9B40,9C40,9D40,9E40, 

               9F40,A040,AA40,AB40,AC40,AD40,AE40,AF40, 

               B040,B140,B240,B340,B440,B540,B640,B740, 

               B840,B940,BA40,BB40,BC40,BD40,BE40,BF40, 

               CA40,CB40,CC40,CD40,CE40,CF40,DA40,DB40, 

               DC40,DD40,DE40,DF40,E140,EA40,EB40,EC40, 

               ED40,EE40,EF40,FA40,FB40,FC40,FD40,FE40, 

               FF40) 

/* 

//BUILDCSV EXEC PGM=SORT 

//SORTOUT  DD SYSOUT=* 

//SYSOUT   DD SYSOUT=* 

//SORTIN   DD DISP=OLD,DSN=*.CONVERT.SORTOUT 

//SYSIN    DD * 

********************************************************************** 

* Alle Inputzeilen mit "*" an erster Stelle und Tabellenüberschrift 

* ignorieren. 

********************************************************************** 

 OMIT COND=(1,1,CH,EQ,C'*',OR,30,5,FS,NE,NUM) 

 OPTION COPY 

*********************************************************************** 

* *---+----1----+----2----+----3----+----4----+----5----+----6----+---- 

* Adebar       Hunziker        07041 Schlangenhausen An der Feder 287   

* Don Hans     von Burger      07066 Karlskühe       Glocken Weg 23 A   

*********************************************************************** 

 INREC  BUILD=(1,13,SQZ=(SHIFT=LEFT,MID=C'~'), 

               14,16,SQZ=(SHIFT=LEFT,MID=C'~'), 

               30,6, 

               36,16,SQZ=(SHIFT=LEFT,MID=C'~'), 

               52,29,SQZ=(SHIFT=LEFT,MID=C'~'), 

               81,10) 

*********************************************************************** 

* Aufbereitung im INREC - Statement: 

* - Felder werden linksbündig gestaucht. 

* - Bei mehreren Wörtern werden ALLE Blanks entfernt. 

* - Mit "MID" wird eine Zeichenkette angegeben die zwischen den 

*   Wörtern eingesetzt wird. 

* *---+----1----+----2----+----3----+----4----+----5----+----6----+---- 

* Adebar       Hunziker        07041 Schlangenhausen An~der~Feder~287   

* Don~Hans     von~Burger      07066 Karlskühe       Glocken~Weg~23~A   

*********************************************************************** 

   OUTREC BUILD=(1,90,SQZ=(SHIFT=LEFT,MID=C';')) 

*********************************************************************** 

* Aufbereitung im OUTREC - Statement: 

* - Die gesamte Zeile wird linksbündig gestaucht mit ";" als 

*   Worttrennung 

* *---+----1----+----2----+----3----+----4----+----5----+----6----+---- 

* Adebar;Hunziker;07041;Schlangenhausen;An~der~Feder~287;05.10.1953 

* Don~Hans;von~Burger;07066;Karlskühe;Glocken~Weg~23~A;24.02.1983 

*********************************************************************** 

   OUTFIL FINDREP=(IN=C'~',OUT=C' ') 

*********************************************************************** 

* Aufbereitung im OUTFIL - Statement: 

* - Alle "~" werden durch einen Blank ersetzt 

* *---+----1----+----2----+----3----+----4----+----5----+----6----+---- 

* Adebar;Hunziker;07041;Schlangenhausen;An der Feder 287;05.10.1953     

* Don Hans;von Burger;07066;Karlskühe;Glocken Weg 23 A;24.02.1983       

*********************************************************************** 

/* 

7.12.6.12. Feldformatierung 

//FORMAT   EXEC PGM=SORT 

//AUFGABE  DD * 

1. Die Daten von Spalte 62 bis 70 sind Cent-Beträge. Diese sollen 

   rechtsbündig auf Spalte 70 ausgerichtet werden. 

2. Die Daten (Preisfeld) sollen linksseitig mit führenden Nullen 

   aufgefüllt werden 

3. An Stelle des Euro-Zeichens (€) soll die Zeichenfolge Cent stehen. 

 

*---+----1----+----2----+----3----+----4----+----5----+----6----+----7- 

05 11111 001000 L              Bier                     €    10000 

05 11111 001000 L              Bier                     Cent 000010000 

/* 

//SYSOUT   DD SYSOUT=* 

//SORTOUT  DD SYSOUT=* 

//SORTIN   DD DISP=SHR,DSN=SP1.EDU.DATA(SORTIN#6) 

//SYSIN    DD * 

********************************************************************** 

* Alle Inputzeilen mit "*" an erster Stelle ignorieren. 

********************************************************************** 

   OMIT COND=(1,1,CH,EQ,C'*') 

   OPTION COPY 

********************************************************************** 

* Mit FINDREP kann man Zeichenfolgen ersetzen. (siehe Manual Seite 142) 

********************************************************************** 

   INREC   FINDREP=(IN=C'€',OUT=C'Cent') 

********************************************************************** 

* Mit einer Maske kann man die Darstellung von numerischen Feldern 

* bestimmen. (siehe Manual Seite 262) 

********************************************************************** 

   OUTREC  BUILD=(1,61,62,9,UFF,M11,LENGTH=9) 

/* 

7.12.6.13. Parsing der Eingabesätze 

//PARSE    EXEC PGM=SORT 

//AUFGABE  DD * 

An beliebiger zwischen 2 Eye-Catchern (~) liegt eine IBAN. Aufgabe ist, 

die IBAN zu finden und ausgabesätze zu schreiben, bei denen die IBAN 

auf Stelle 1 steht und ab Stelle 30 BLZ und Kontonummer ausgegeben wird. 

Eingabeformat (Beispiel) 

Blurp Sülz ~DE42285622970218951800~Gemischtwarenhandel 

 

Format der Ausgabe (bezogen auf das Beispiel der Eingabe) 

DE42 2856 2297 0218 9518 00  BLZ(285 622 97) KtNr: 281951800 

 

/* 

//SORTIN   DD DISP=SHR,DSN=SP1.EDU.DATA(SORTIN#7) 

//SORTOUT  DD SYSOUT=* 

//SYSOUT   DD SYSOUT=* 

//SYSIN    DD * 

********************************************************************** 

* Alle Inputzeilen mit "*" an erster Stelle ignorieren. 

********************************************************************** 

  OMIT COND=(1,1,CH,EQ,C'*') 

  OPTION COPY 

********************************************************************** 

* Entfernen aller Leerzeichen mit FINDREP. (siehe Manual Seite 142) 

********************************************************************** 

 INREC  FINDREP=(IN=C' ',OUT=C'') 

********************************************************************** 

* IBAN zwichen den beiden Tilden in die Variable %00 stellen. 

********************************************************************** 

 OUTREC PARSE=(%00=(STARTAFT=C'~',ENDBEFR=C'~',FIXLEN=22), 

********************************************************************** 

* Neue Startposition -23 

********************************************************************** 

                 %=(SUBPOS=23), 

********************************************************************** 

* IBAN in Blöcke aufteilen 

********************************************************************** 

               %01=(FIXLEN=4),   * IBAN Block 1 (Länge 4) 

               %02=(FIXLEN=4),   * IBAN Block 2 (Länge 4) 

               %03=(FIXLEN=4),   * IBAN Block 3 (Länge 4) 

               %04=(FIXLEN=4),   * IBAN Block 4 (Länge 4) 

               %05=(FIXLEN=4),   * IBAN Block 5 (Länge 4) 

               %06=(FIXLEN=2),   * IBAN Block 6 (Länge 2) 

                 %=(SUBPOS=18),  * Neue Startposition -18 

********************************************************************** 

* BLZ  in Blöcke aufteilen 

********************************************************************** 

               %11=(FIXLEN=3),   * BLZ Block 1 (Länge 3) 

               %12=(FIXLEN=3),   * BLZ Block 2 (Länge 3) 

               %13=(FIXLEN=2),   * BLZ Block 3 (Länge 2) 

               %14=(FIXLEN=10)), * Kontonummer 

********************************************************************** 

* Ausgabezeile bilden. Bei Kontonummer führende Nullen nicht 

* darstellen. 

********************************************************************** 

        BUILD=(%01,C' ',%02,C' ',%03,C' ',%04,C' ',%05,C' ',%06,C'  ', 

               C'BLZ(',%11,C' ',%12,C' ',%13, 

               C') KtNr:',%14,UFF,M0) 

/* 

7.12.6.14. Umblocken von Daten 

Die Eingabedatei weist eine Blocksize von 270 Byte auf und soll auf 27900 Bytes umgeblockt wer-
den. Klingt sehr einfach, aber die meisten Programme können das nicht. 

//CLEANUP  EXEC PGM=EQQDELDS 

//SYSPRINT DD SYSOUT=* 

//REBLOCK  EXEC PGM=SORT 

//AUFGABE  DD * 

Siehe Seite 60 "Table 9. DCB Subparameters Used by DFSORT" 

/* 

//SYSOUT   DD SYSOUT=* 

//SORTIN   DD DISP=SHR,DSN=SP1.EDU90PS.DATA 

//SORTOUT  DD DISP=(,CATLG), 

//            DSN=&SYSUID..REBLOCKT.EDU90PS.DATA, 

//            SPACE=(CYL,1),   Oder mehr 

//            BLKSIZE=27900 

//SYSIN    DD * 

 OPTION COPY 

/* 

7.12.7. CUNMRCSM (Mails vom Host) 

In vielen Unternehmen gibt es die Möglichkeit, Mails vom Host zu verschicken. In den meisten Fällen 
sind das irgendwelche REXXe, zusammengetragen auf Guides oder gefunden im Internet. Sie alle 
basieren auf SMTP und würde man nur das benutzen, was ursprünglich vorgesehen war, wäre die 
Welt auch in Ordnung. 

Leider wurden diese Programme im Lauf der Zeit immer mehr verschlimmbessert (eingebettetes 
HTML, Attachments, etc.), sodass in den meisten Unternehmen hierfür Routinen im Einsatz sind, 
von denen niemand weiß, ob sie beim nächsten zOS-Releasewechsel noch funktionieren. 

Seit zOS 2.10 gibt es ein offizielles Programm der IBM, welches alles, was das Herz begehrt, kann. 
Und es wird von der IBM gewartet. All denen, die es noch nicht benutzen, möchte ich es wärmstens 
ans Herz legen. Trennt euch von Selbstgestrickten, wenn es etwas vergleichbares vom Hersteller 
gibt. 

//* Mails von Host mit 

//* 

//* ° Eingebettetem Text 

//* ° Eingebettetem Text aus Datei 

//* ° Eingebettetem HTML aus Datei 

//* ° Attachments als EBCIC-Datei 

//* ° Attachments als BIN-Datei 

//* 

//MAIL     EXEC PGM=CUNMRCSM,REGION=0M 

//SNAPDUMP DD DUMMY 

//SYSOUT   DD SYSOUT=* 

//MIMOUT   DD SYSOUT=(B,TCPSMTP) 

//HINWEISE DD * Wird nicht benutzt - ist Kommentar 

CRLF=2     * Einfügen von Zeilenschaltern (CRLF) beim Lesen jeden Satzes 

           * Verschiedene Werte möglich 

SUBJ=text  * Betreff-Zeile der Mail 

RPLY=adr   * Empfänger bei <Antworten> 

RECP=adr   * Mailempfänger, kann beliebig oft kodiert sein 

COPY=adr   * Mailempfänger, in Kopie 

BCPY=adr   * Mailempfänger, in Blindkopie 

//PARMDD   DD * 

EHLO 

HOST=MVS1.MAILING.COMPANY.DE 

SUBJ=Demo für Mails mit Attachment via CUNMRCSM 

ORIG=KW@SPGMBH.DE.DE 

RPLY=REXXIKON@SPGMBH.DE.DE 

RECP=KW@SPGMBH.DE.DE 

 

DATA=&TEXT 

DATA=&IMBED1  TXT 

DATA=&IMBED2  TXT 

CRLF=2 

DATA=&ATT1    TXT Adress-Liste.TXT 

DATA=&ATT2    BIN Rexx-Styleguide.DOC 

//TEXT     DD * 

Sie haben Post mit Anhang 

 

 

Schöne Grüße vom 

Hosti-Auto-Mail-o-mat 

 

//IMBED1   DD DISP=SHR,DSN=SP1.EDU.DATA(SEND2) 

//IMBED2   DD DISP=SHR,DSN=SP1.EDU.DATA(SEND3) 

//ATT1     DD DISP=SHR,DSN=SP1.EDU.DATA(MAILATT1) 

//ATT2     DD DISP=SHR,DSN=SP1.RXSTYLE.DATA     Ist Binary 

8. Datenstrukturen der Beispiele 

Wann immer Beispieldaten benötigt werden, benützen die Musterlösungen Member der Datei 
SP1.EDU.DATA (PO/E, FB132). Damit nachvollziehbar ist, auf welcher Datenbasis die Lösungen 
funktionieren, werden im Anschluss die Aufbauten der Member gezeigt. 

8.1.1. ADR1, ADRTEST, ADRSORT, MAILATT1, SORTIN#1 

Alle drei sind vom Aufbau her gleich. Sie unterscheiden sich lediglich in der Anzahl der Datensätze 
und der Sortierung: 

----+----1----+----2----+----3----+----4----+----5----+----6----+----7--- 

Friederich   von Blumendraht 01101 Gronau          Dohlenweg 578 

Katharina    von Draußenrein 01139 Oberarm         Akneweg 236 

August       von Stress      01388 Quellcode       Ersatzteilplatz 578 

Utz          Albers          01510 Langarm         Murrbastrog 406 

 

-+----8----+----9----+----0----+----1----+----2----+----3-- 

       04.08.1990 

       09.10.1939 

       23.09.1972 

       30.01.1948 

8.1.2. ADRDB2 

----+----1----+----2----+----3----+----4----+----5----+----6----+----7 

Vorname     !Nachname       !PLZ  !Ort            !Straße 

Gilda       !Grinser        !01101!Bernbach       !Reeperbahn 158 

Jonathan    !Joystick       !01139!Untergrombach  !Beingasse 539 

Xaver       !Idefix         !01388!Himmelreich    !Kugelgasse 988          

Barny       !Iltis          !01510!Vandalistenheim!Piste 758 

Zenzi       !Video          !01775!Hummelflug     !Ovale Ecke 734 

 

----+----8----+----9----+----0----+----1----+----2----+----3-- 

    !Geb.-Datum!Pseudonym     !Hobby             !Abteilung    

    !10.02.1963!Rumbalotte    !Essen & Schlafen  !Vertrieb     

    !02.01.1954!Sesselpupser  !Essen & Trinken   !Einkauf      

    !19.08.1934!              !                  !             

    !05.10.1953!Rübe          !                  !Einkauf      

    !24.02.1934!              !                  !             

8.1.3. ITPREIS 

----+----1----+----2----+----3----+----4----+----5----+----6----+----7— 

* Die Preisangaben sind in Cent 

010040 Bohnenkraut                                 Cent  0000060 

021145 Spannungsabfall                             Cent  0001250 

014080 Bitverschnitt                               Cent  0000270 

051099 Blockluecken                                Cent  0010780 

050520 Leitungsprotokolle                          Cent  0036050 

019010 Bierdeckel                                  Cent  0003058 

8.1.4. SEND2 

      __                 ___                 __             

  ___(  \             .--   --.             /  )___         

 (___    \____       /  (O O)  \       ____/    ___)        

(___          .-----(     )     )-----.          ___)       

 (___    ____________\  \___/  /____________    ___)        

   (____/              -.___.-              \____)          

                                                        

8.1.5. SEND3 

                         ___                                

                        .   .                               

                       ( o o )                              

+-----------------.oooO--(_)--Oooo.-----------------+       

|                                                   |       

| Hallo Heinzi - Du bist der Beste und der Schönste |       

|                                                   |       

|                   .oooO                           |       

|                   (   )   Oooo.                   |       

+--------------------\ (----(   )-------------------+       

                      \_)    ) /                            

                            (_/                             

                                                            

8.1.6. SORTIN#2 

* Eingabe für Sortaufgabe.                                                     

*                                                                              

* Ziel der Verarbeitung:                                                       

*                                                                              

* Es gibt 3 Members, welche zusammengemischt werden sollen.                    

* SORTIN#2 (dieses) enthält eine UserID und die dazugehörende Abteilung        

* SORTIN#3 enthält eine UserID und den dazugehörenden Vor- und Nachnamen       

* SORTIN#4 enthält eine UserID und die dazugehörende Telefonnummer             

*                                                                              

* Aus diesen drei Eingaben sollen Ausgabesätze gebildet werden, in denen       

* UserID - Abteilung - Vor- und Nachname und die Telefonnummer steht.          

*                                                                              

USR0031 PLSMFS                                                                 

USR4711 PLSATM                                                                 

USR0815 PLSDBM                                                                 

USR7022 PLSWUS 

8.1.7. SORTIN#3 

USR0031 August Nieselprim         

USR4711 Barny Geröllheimer        

USR0815 Maurizio Krawallo         

USR7022 Kreszenzia Hosndobl 

8.1.8. SORTIN#4 

USR0031 (07083) 52 49 65       

USR4711 (0721) 4004 2066       

USR0815 (0176) 5624 4567       

USR7022 (089)  1234 5678 

8.1.9. SORTIN#5 

Wie ADR1. Allerdings liegt im ersten Datensatz an Stelle 1 ein "20"x. Dieses Zeichen ist nicht sicht-
bar (sieht aus, als stünde dort ein Blank ("40"x). 

8.1.10. SORTIN#6 

* Eingabe für Sortaufgabe.                                                                                                           

*                                                                                                                                    

* Ziel der Verarbeitung:                                                                                                             

*                                                                                                                                    

* 1. Die Daten von Spalte 62 bis 70 sind Cent-Beträge. Diese sollen                                                                  

*    rechtsbündig auf Spalte 70 ausgerichtet wer-

den.                                                                                 

* 2. Die Daten (Preisfeld) sollen linksseitig mit führenden Nullen                                                                   

*    aufgefüllt wer-

den.                                                                                                              

* 3. An Stelle des Euro-Zeichens (¤) soll die Zeichenfolge Cent stehen.                                                              

*---+----1----+----2----+----3----+----4----+----5----+----6----+----7--- 

05 11111 001000 L              Bier                     €    10000 

05 11111 001001 L              Bier                     €    10000 

05 45810 135750 S              Lufthaken                €    33000 

01 09715 002350 S              Brogramierer             €    75 

01 50040 000230 K  Hier steht  Bohnenkraut              €    60 

01 70730 120500 S              Pappnasen                €    13604 

8.1.11. SORTIN#7 

* Eingabe für Sortaufgabe. 

* 

* Ziel der Verarbeitung: 

* 

* An wahlfreier Stelle steht eine IBAN, welche zwischen zwei Eyecathern 

* liegt (~). Aufgabe ist, die IBAN zu finden, Ausgabesätze zu schreiben, 

* bei denen die IBAN auf Stelle 1 steht und ab Stelle 30 BLZ und 

* Kontonummer ausgegeben wird. 

* 

* Format der Ausgabe (bezogen auf den ersten Satz): 

* DE42 2856 2297 0218 9518 00  BLZ(285 622 97) KtNr: 281951800 

* 

